



LÊ MINH HOÀNG



**GIẢI THUẬT
&
LẬP TRÌNH**

(A.K.A DSAP Textbook)

Đại học Sư phạm Hà Nội, 1999-2002

*Try not to become a man of success
but rather to become a man of value.*

Albert Einstein

MỤC LỤC

PHẦN 1. BÀI TOÁN LIỆT KÊ	1
§1. NHẮC LẠI MỘT SỐ KIẾN THỨC ĐẠI SỐ TỔ HỢP	2
1.1. CHÍNH HỢP LẬP	2
1.2. CHÍNH HỢP KHÔNG LẬP	2
1.3. HOÁN VỊ	2
1.4. TỔ HỢP	3
§2. PHƯƠNG PHÁP SINH (GENERATION)	4
2.1. SINH CÁC DÃY NHỊ PHÂN ĐỘ DÀI N	5
2.2. LIỆT KÊ CÁC TẬP CON K PHẦN TỬ	6
2.3. LIỆT KÊ CÁC HOÁN VỊ	8
§3. THUẬT TOÁN QUAY LUI	12
3.1. LIỆT KÊ CÁC DÃY NHỊ PHÂN ĐỘ DÀI N	12
3.2. LIỆT KÊ CÁC TẬP CON K PHẦN TỬ	13
3.3. LIỆT KÊ CÁC CHÍNH HỢP KHÔNG LẬP CHẬP K	15
3.4. BÀI TOÁN PHÂN TÍCH SỐ	17
3.5. BÀI TOÁN XẾP HẬU	19
§4. KỸ THUẬT NHÁNH CẬN	24
4.1. BÀI TOÁN TỐI ƯU	24
4.2. SỰ BÙNG NÓ TỔ HỢP	24
4.3. MÔ HÌNH KỸ THUẬT NHÁNH CẬN	24
4.4. BÀI TOÁN NGƯỜI DU LỊCH	25
4.5. DÃY ABC	27
PHẦN 2. CẤU TRÚC DỮ LIỆU VÀ GIẢI THUẬT	33
§1. CÁC BƯỚC CƠ BẢN KHI TIẾN HÀNH GIẢI CÁC BÀI TOÁN TIN HỌC	34
1.1. XÁC ĐỊNH BÀI TOÁN	34
1.2. TÌM CẤU TRÚC DỮ LIỆU BIỂU DIỄN BÀI TOÁN	34
1.3. TÌM THUẬT TOÁN	35
1.4. LẬP TRÌNH	37
1.5. KIỂM THỦ	37
1.6. TỐI ƯU CHƯƠNG TRÌNH	38
§2. PHÂN TÍCH THỜI GIAN THỰC HIỆN GIẢI THUẬT	40
2.1. GIỚI THIỆU	40
2.2. CÁC KÝ PHÁP ĐỀ ĐÁNH GIÁ ĐỘ PHÚC TẠP TÍNH TOÁN	40
2.3. XÁC ĐỊNH ĐỘ PHÚC TẠP TÍNH TOÁN CỦA GIẢI THUẬT	42
2.4. ĐỘ PHÚC TẠP TÍNH TOÁN VỚI TÌNH TRẠNG DỮ LIỆU VÀO	45
2.5. CHI PHÍ THỰC HIỆN THUẬT TOÁN	46

§3. ĐỆ QUY VÀ GIẢI THUẬT ĐỆ QUY	50
3.1. KHÁI NIỆM VỀ ĐỆ QUY	50
3.2. GIẢI THUẬT ĐỆ QUY	50
3.3. VÍ DỤ VỀ GIẢI THUẬT ĐỆ QUY	51
3.4. HIỆU LỰC CỦA ĐỆ QUY	55
§4. CẤU TRÚC DỮ LIỆU BIỂU DIỄN DANH SÁCH.....	58
4.1. KHÁI NIỆM DANH SÁCH	58
4.2. BIỂU DIỄN DANH SÁCH TRONG MÁY TÍNH	58
§5. NGĂN XẾP VÀ HÀNG ĐỢI	64
5.1. NGĂN XẾP (STACK).....	64
5.2. HÀNG ĐỢI (QUEUE).....	66
§6. CÂY (TREE).....	70
6.1. ĐỊNH NGHĨA	70
6.2. CÂY NHỊ PHÂN (BINARY TREE)	71
6.3. BIỂU DIỄN CÂY NHỊ PHÂN	73
6.4. PHÉP DUYỆT CÂY NHỊ PHÂN	74
6.5. CÂY K_PHÂN	76
6.6. CÂY TỔNG QUÁT	77
§7. KÝ PHÁP TIỀN TỐ, TRUNG TỐ VÀ HẬU TỐ	79
7.1. BIÊU THỨC DƯỚI DẠNG CÂY NHỊ PHÂN	79
7.2. CÁC KÝ PHÁP CHO CÙNG MỘT BIÊU THỨC.....	79
7.3. CÁCH TÍNH GIÁ TRỊ BIÊU THỨC	79
7.4. CHUYỂN TỪ DẠNG TRUNG TỐ SANG DẠNG HẬU TỐ.....	83
7.5. XÂY DỰNG CÂY NHỊ PHÂN BIỂU DIỄN BIÊU THỨC	86
§8. SẮP XẾP (SORTING)	88
8.1. BÀI TOÁN SẮP XẾP	88
8.2. THUẬT TOÁN SẮP XẾP KIỀU CHỌN (SELECTIONSORT)	90
8.3. THUẬT TOÁN SẮP XẾP NỒI BỌT (BUBBLESORT).....	91
8.4. THUẬT TOÁN SẮP XẾP KIỀU CHÈN (INSERTIONSORT)	91
8.5. SẮP XẾP CHÈN VỚI ĐỘ DÀI BUỚC GIẢM DÀN (SHELLSORT)	93
8.6. THUẬT TOÁN SẮP XẾP KIỀU PHÂN ĐOẠN (QUICKSORT)	94
8.7. THUẬT TOÁN SẮP XẾP KIỀU VUN ĐÓNG (HEAPSORT)	101
8.8. SẮP XẾP BẰNG PHÉP ĐÉM PHÂN PHỐI (DISTRIBUTION COUNTING).....	104
8.9. TÍNH ÔN ĐỊNH CỦA THUẬT TOÁN SẮP XẾP (STABILITY)	105
8.10. THUẬT TOÁN SẮP XẾP BẰNG CƠ SỐ (RADIX SORT).....	106
8.11. THUẬT TOÁN SẮP XẾP TRỘN (MERGESORT).....	111
8.12. CÀI ĐẶT	114
8.13. ĐÁNH GIÁ, NHẬN XÉT	122
§9. TÌM KIẾM (SEARCHING)	126
9.1. BÀI TOÁN TÌM KIẾM	126
9.2. TÌM KIẾM TUẦN TỰ (SEQUENTIAL SEARCH)	126
9.3. TÌM KIẾM NHỊ PHÂN (BINARY SEARCH).....	126
9.4. CÂY NHỊ PHÂN TÌM KIẾM (BINARY SEARCH TREE - BST).....	127

9.5. PHÉP BĂM (HASH)	132
9.6. KHOÁ SỐ VỚI BÀI TOÁN TÌM KIẾM	132
9.7. CÂY TÌM KIẾM SỐ HỌC (DIGITAL SEARCH TREE - DST)	133
9.8. CÂY TÌM KIẾM CƠ SỐ (RADIX SEARCH TREE - RST)	136
9.9. NHỮNG NHẬN XÉT CUỐI CÙNG	141

PHẦN 3. QUY HOẠCH ĐỘNG 143

§1. CÔNG THỨC TRUY HỒI	144
1.1. VÍ DỤ	144
1.2. CÀI TIẾN THỨ NHẤT	145
1.3. CÀI TIẾN THỨ HAI	147
1.4. CÀI ĐẶT ĐỀ QUY	147
§2. PHƯƠNG PHÁP QUY HOẠCH ĐỘNG	149
2.1. BÀI TOÁN QUY HOẠCH	149
2.2. PHƯƠNG PHÁP QUY HOẠCH ĐỘNG	149
§3. MỘT SỐ BÀI TOÁN QUY HOẠCH ĐỘNG	153
3.1. DÃY CON ĐƠN ĐIỆU TĂNG DÀI NHẤT	153
3.2. BÀI TOÁN CÁI TÚI	158
3.3. BIÊN ĐỒI XÂU	160
3.4. DÃY CON CÓ TÔNG CHIA HẾT CHO K	164
3.5. PHÉP NHÂN TỔ HỢP DÃY MA TRẬN	169
3.6. BÀI TẬP LUYỆN TẬP	172

PHẦN 4. CÁC THUẬT TOÁN TRÊN ĐỒ THỊ 177

§1. CÁC KHÁI NIỆM CƠ BẢN	178
1.1. ĐỊNH NGHĨA ĐỒ THỊ (GRAPH)	178
1.2. CÁC KHÁI NIỆM	179
§2. BIỂU DIỄN ĐỒ THỊ TRÊN MÁY TÍNH	181
2.1. MA TRẬN KÈ (ADJACENCY MATRIX)	181
2.2. DANH SÁCH CẠNH (EDGE LIST)	182
2.3. DANH SÁCH KÈ (ADJACENCY LIST)	183
2.4. NHẬN XÉT	184
§3. CÁC THUẬT TOÁN TÌM KIẾM TRÊN ĐỒ THỊ	186
3.1. BÀI TOÁN	186
3.2. THUẬT TOÁN TÌM KIẾM THEO CHIỀU SÂU (DEPTH FIRST SEARCH)	187
3.3. THUẬT TOÁN TÌM KIẾM THEO CHIỀU RỘNG (BREADTH FIRST SEARCH)	189
3.4. ĐỘ PHÚC TẠP TÍNH TOÁN CỦA BFS VÀ DFS	192
§4. TÍNH LIÊN THÔNG CỦA ĐỒ THỊ	193
4.1. ĐỊNH NGHĨA	193
4.2. TÍNH LIÊN THÔNG TRONG ĐỒ THỊ VÔ HƯỚNG	194

4.3. ĐỒ THỊ ĐÀY ĐỦ VÀ THUẬT TOÁN WARSHALL	194
4.4. CÁC THÀNH PHẦN LIÊN THÔNG MẠNH	197
§5. VÀI ÚNG DỤNG CỦA DFS và BFS	208
5.1. XÂY DỰNG CÂY KHUNG CỦA ĐỒ THỊ.....	208
5.2. TẬP CÁC CHU TRÌNH CƠ SỞ CỦA ĐỒ THỊ.....	211
5.3. BÀI TOÁN ĐỊNH CHIỀU ĐỒ THỊ	211
5.4. LIỆT KÊ CÁC KHỚP VÀ CÀU CỦA ĐỒ THỊ.....	215
§6. CHU TRÌNH EULER, ĐƯỜNG ĐI EULER, ĐỒ THỊ EULER	219
6.1. BÀI TOÁN 7 CÁI CẦU	219
6.2. ĐỊNH NGHĨA.....	219
6.3. ĐỊNH LÝ	219
6.4. THUẬT TOÁN FLEURY TÌM CHU TRÌNH EULER.....	220
6.5. CÀI ĐẶT	221
6.6. THUẬT TOÁN TỐT HƠN.....	223
§7. CHU TRÌNH HAMILTON, ĐƯỜNG ĐI HAMILTON, ĐỒ THỊ HAMILTON	226
7.1. ĐỊNH NGHĨA.....	226
7.2. ĐỊNH LÝ	226
7.3. CÀI ĐẶT	227
§8. BÀI TOÁN ĐƯỜNG ĐI NGẮN NHẤT	231
8.1. ĐỒ THỊ CÓ TRỌNG SỐ.....	231
8.2. BÀI TOÁN ĐƯỜNG ĐI NGẮN NHẤT	231
8.3. TRƯỜNG HỢP ĐỒ THỊ KHÔNG CÓ CHU TRÌNH ÂM - THUẬT TOÁN FORD BELLMAN	233
8.4. TRƯỜNG HỢP TRỌNG SỐ TRÊN CÁC CUNG KHÔNG ÂM - THUẬT TOÁN DIJKSTRA	235
8.5. THUẬT TOÁN DIJKSTRA VÀ CÁU TRÚC HEAP	238
8.6. TRƯỜNG HỢP ĐỒ THỊ KHÔNG CÓ CHU TRÌNH - SẮP XẾP TÔ PÔ.....	241
8.7. ĐƯỜNG ĐI NGẮN NHẤT GIỮA MỌI CẶP ĐỈNH - THUẬT TOÁN FLOYD.....	244
8.8. NHẬN XÉT	246
§9. BÀI TOÁN CÂY KHUNG NHỎ NHẤT	251
9.1. BÀI TOÁN CÂY KHUNG NHỎ NHẤT	251
9.2. THUẬT TOÁN KRUSKAL (JOSEPH KRUSKAL - 1956)	251
9.3. THUẬT TOÁN PRIM (ROBERT PRIM - 1957)	256
§10. BÀI TOÁN LUỒNG CỰC ĐẠI TRÊN MẠNG	260
10.1. CÁC KHÁI NIỆM	260
10.2. MẠNG THĂNG DƯ VÀ ĐƯỜNG TĂNG LUỒNG	263
10.3. THUẬT TOÁN FORD-FULKERSON (L.R.FORD & D.R.FULKERSON - 1962)	266
10.4. THUẬT TOÁN PREFLOW-PUSH (GOLDBERG - 1986)	270
10.5. MỘT SỐ MỞ RỘNG	276
§11. BÀI TOÁN TÌM BỘ GHÉP CỰC ĐẠI TRÊN ĐỒ THỊ HAI PHÍA	283
11.1. ĐỒ THỊ HAI PHÍA (BIPARTITE GRAPH)	283
11.2. BÀI TOÁN GHÉP ĐÔI KHÔNG TRỌNG VÀ CÁC KHÁI NIỆM	283
11.3. THUẬT TOÁN ĐƯỜNG MỞ	284
11.4. CÀI ĐẶT	285

§12. BÀI TOÁN TÌM BỘ GHÉP CỰC ĐẠI VỚI TRỌNG SÓ CỰC TIẾU TRÊN ĐỒ THỊ HAI PHÍA - THUẬT TOÁN HUNGARI	291
12.1. BÀI TOÁN PHÂN CÔNG	291
12.2. PHÂN TÍCH	291
12.3. THUẬT TOÁN	292
12.4. BÀI TOÁN TÌM BỘ GHÉP CỰC ĐẠI VỚI TRỌNG SÓ CỰC ĐẠI TRÊN ĐỒ THỊ HAI PHÍA.....	301
12.5. NÂNG CẤP	301
§13. BÀI TOÁN TÌM BỘ GHÉP CỰC ĐẠI TRÊN ĐỒ THỊ.....	307
13.1. CÁC KHÁI NIỆM	307
13.2. THUẬT TOÁN EDMONDS (1965)	308
13.3. THUẬT TOÁN LAWLER (1973).....	310
13.4. CÀI ĐẶT	312
13.5. ĐỘ PHÚC TẠP TÍNH TOÁN.....	316
TÀI LIỆU ĐỌC THÊM	319

HÌNH VẼ

Hình 1: Cây tìm kiếm quay lui trong bài toán liệt kê dãy nhị phân	13
Hình 2: Xếp 8 quân hậu trên bàn cờ 8x8	19
Hình 3: Đường chéo ĐB-TN mang chỉ số 10 và đường chéo ĐN-TB mang chỉ số 0.....	20
Hình 4: Lưu đồ thuật giải (Flowchart).....	36
Hình 5: Ký pháp Θ lớn, O lớn và Ω lớn	41
Hình 6: Tháp Hà Nội	54
Hình 7: Cấu trúc nút của danh sách nối đơn	59
Hình 8: Danh sách nối đơn	59
Hình 9: Cấu trúc nút của danh sách nối kép	61
Hình 10: Danh sách nối kép.....	61
Hình 11: Danh sách nối vòng một hướng	61
Hình 12: Danh sách nối vòng hai hướng	62
Hình 13: Dùng danh sách vòng mô tả Queue	67
Hình 14: Di chuyển toa tàu	69
Hình 15: Di chuyển toa tàu (2)	69
Hình 16: Cây.....	70
Hình 17: Mức của các nút trên cây	71
Hình 18: Cây biểu diễn biểu thức	71
Hình 19: Các dạng cây nhị phân suy biến.....	72
Hình 20: Cây nhị phân hoàn chỉnh và cây nhị phân đầy đủ.....	72
Hình 21: Đánh số các nút của cây nhị phân đầy đủ để biểu diễn bằng mảng	73
Hình 22: Nhược điểm của phương pháp biểu diễn cây bằng mảng	73
Hình 23: Cấu trúc nút của cây nhị phân.....	74
Hình 24: Biểu diễn cây bằng cấu trúc liên kết	74
Hình 25: Đánh số các nút của cây 3_phân để biểu diễn bằng mảng	76
Hình 26: Biểu diễn cây tổng quát bằng mảng.....	77
Hình 27: Cấu trúc nút của cây tổng quát.....	78
Hình 28: Biểu thức dưới dạng cây nhị phân	79
Hình 29: Vòng lặp trong của QuickSort	95
Hình 30: Trạng thái trước khi gọi đệ quy	96
Hình 31: Heap.....	102
Hình 32: Vun đống	102
Hình 33: Đảo giá trị k[1] cho k[n] và xét phần còn lại	103
Hình 34: Vun phần còn lại thành đống rồi lại đảo trị k[1] cho k[n-1]	103
Hình 35: Đánh số các bit	106
Hình 36: Thuật toán sắp xếp trộn.....	111

Hình 37: Máy Pentium 4, 3.2GHz, 2GB RAM tỏ ra chậm chạp khi sắp xếp 10^8 khoá $\in [0..7.10^7]$ cho dù những thuật toán sắp xếp tốt nhất đã được áp dụng	123
Hình 38: Cây nhị phân tìm kiếm	128
Hình 39: Xóa nút lá ở cây BST	129
Hình 40. Xóa nút chỉ có một nhánh con trên cây BST	130
Hình 41: Xóa nút có cả hai nhánh con trên cây BST thay bằng nút cực phải của cây con trái.....	130
Hình 42: Xóa nút có cả hai nhánh con trên cây BST thay bằng nút cực trái của cây con phải.....	130
Hình 43: Đánh số các bit	133
Hình 44: Cây tìm kiếm số học	133
Hình 45: Cây tìm kiếm cơ sở	136
Hình 46: Với độ dài dãy bit $z = 3$, cây tìm kiếm cơ sở gồm các khoá 2, 4, 5 và sau khi thêm giá trị 7	137
Hình 47: RST chứa các khoá 2, 4, 5, 7 và RST sau khi loại bỏ giá trị 7	138
Hình 48: Cây tìm kiếm cơ sở a) và Trie tìm kiếm cơ sở b).....	140
Hình 49: Hàm đệ quy tính số Fibonacci	151
Hình 50: Tính toán và truy vết	154
Hình 51: Truy vết	163
Hình 52: Ví dụ về mô hình đồ thị	178
Hình 53: Phân loại đồ thị	179
Hình 54	182
Hình 55	183
Hình 56: Đồ thị và đường đi	186
Hình 57: Cây DFS	189
Hình 58: Cây BFS	192
Hình 59: Đồ thị G và các thành phần liên thông G1, G2, G3 của nó	193
Hình 60: Khớp và cầu	193
Hình 61: Liên thông mạnh và liên thông yếu	194
Hình 62: Đồ thị dày đủ	195
Hình 63: Đơn đồ thị vô hướng và bao đóng của nó	195
Hình 64: Ba dạng cung ngoài cây DFS	199
Hình 65: Thuật toán Tarjan “bé” cây DFS	201
Hình 66: Đánh số lại, đảo chiều các cung và duyệt BFS với cách chọn các đỉnh xuất phát ngược lại với thứ tự duyệt xong (thứ tự 11, 10... 3, 2, 1).....	206
Hình 67: Đồ thị G và một số ví dụ cây khung T1, T2, T3 của nó	210
Hình 68: Cây khung DFS (a) và cây khung BFS (b) (Mũi tên chỉ chiều đi thăm các đỉnh)	210
Hình 69: Phép định chiều DFS	213
Hình 70: Phép đánh số và ghi nhận cung ngược lên cao nhất	215
Hình 71: Mô hình đồ thị của bài toán bảy cái cầu	219
Hình 72	220
Hình 73	220
Hình 74	226

Hình 75: Phép đánh lại chỉ số theo thứ tự tópô.....	241
Hình 76: Hai cây gốc r_1 và r_2 và cây mới khi hợp nhất chúng.....	252
Hình 77: Mạng với các khả năng thông qua (1 phát, 6 thu) và một luồng của nó với giá trị 7	260
Hình 78: Mạng G và mạng thặng dư G_f tương ứng (ký hiệu $f[u,v]:c[u,v]$ chỉ luồng $f[u, v]$ và khả năng thông qua $c[u, v]$ trên cung (u, v))	264
Hình 79: Mạng thặng dư và đường tăng luồng	265
Hình 80: Luồng trên mạng G trước và sau khi tăng.....	265
Hình 81: Mạng giả của mạng có nhiều điểm phát và nhiều điểm thu.....	276
Hình 82: Thay một đỉnh u bằng hai đỉnh u_{in}, u_{out}	277
Hình 83: Mạng giả của mạng có khả năng thông qua của các cung bị chặn hai phía	277
Hình 84: Đồ thị hai phía	283
Hình 85: Đồ thị hai phía và bộ ghép M	284
Hình 86: Mô hình luồng của bài toán tìm bộ ghép cực đại trên đồ thị hai phía	288
Hình 87: Phép xoay trọng số cạnh	292
Hình 88: Thuật toán Hungari	295
Hình 89: Cây pha “mọc” lớn hơn sau mỗi lần xoay trọng số cạnh và tìm đường.....	302
Hình 90: Đồ thị G và một bộ ghép M	307
Hình 91: Phép chập Blossom.....	309
Hình 92: Nở Blossom để dò đường xuyên qua Blossom	309

CHƯƠNG TRÌNH

P_1_02_1.PAS * Thuật toán sinh liệt kê các dãy nhị phân độ dài n	6
P_1_02_2.PAS * Thuật toán sinh liệt kê các tập con k phần tử	8
P_1_02_3.PAS * Thuật toán sinh liệt kê hoán vị	9
P_1_03_1.PAS * Thuật toán quay lui liệt kê các dãy nhị phân độ dài n	12
P_1_03_2.PAS * Thuật toán quay lui liệt kê các tập con k phần tử	14
P_1_03_3.PAS * Thuật toán quay lui liệt kê các chỉnh hợp không lặp chập k	16
P_1_03_4.PAS * Thuật toán quay lui liệt kê các cách phân tích số	18
P_1_03_5.PAS * Thuật toán quay lui giải bài toán xếp hậu	21
P_1_04_1.PAS * Kỹ thuật nhánh cận dùng cho bài toán người du lịch	26
P_1_04_2.PAS * Dãy ABC	28
P_2_07_1.PAS * Tính giá trị biểu thức RPN	81
P_2_07_2.PAS * Chuyển biểu thức trung tố sang dạng RPN	84
P_2_08_1.PAS * Các thuật toán sắp xếp	114
P_3_01_1.PAS * Đếm số cách phân tích số n	145
P_3_01_2.PAS * Đếm số cách phân tích số n	146
P_3_01_3.PAS * Đếm số cách phân tích số n	146
P_3_01_4.PAS * Đếm số cách phân tích số n	147
P_3_01_5.PAS * Đếm số cách phân tích số n dùng đệ quy	147
P_3_01_6.PAS * Đếm số cách phân tích số n dùng đệ quy	148
P_3_03_1.PAS * Tìm dãy con đơn điệu tăng dài nhất	154
P_3_03_2.PAS * Cải tiến thuật toán tìm dãy con đơn điệu tăng dài nhất	156
P_3_03_3.PAS * Bài toán cái túi	159
P_3_03_4.PAS * Biến đổi xâu	163
P_3_03_5.PAS * Dãy con có tổng chia hết cho k	165
P_3_03_6.PAS * Dãy con có tổng chia hết cho k	167
P_3_03_7.PAS * Nhân tối ưu dãy ma trận	171
P_4_03_1.PAS * Thuật toán tìm kiếm theo chiều sâu	187
P_4_03_2.PAS * Thuật toán tìm kiếm theo chiều rộng	190
P_4_04_1.PAS * Thuật toán Warshall liệt kê các thành phần liên thông	197
P_4_04_2.PAS * Thuật toán Tarjan liệt kê các thành phần liên thông mạnh	204
P_4_05_1.PAS * Liệt kê các khớp và cầu của đồ thị	216
P_4_06_1.PAS * Thuật toán Fleury tìm chu trình Euler	221
P_4_06_2.PAS * Thuật toán hiệu quả tìm chu trình Euler	224
P_4_07_1.PAS * Thuật toán quay lui liệt kê chu trình Hamilton	227
P_4_08_1.PAS * Thuật toán Ford-Bellman	234
P_4_08_2.PAS * Thuật toán Dijkstra	236
P_4_08_3.PAS * Thuật toán Dijkstra và cấu trúc Heap	239

P_4_08_4.PAS * Đường đi ngắn nhất trên đồ thị không có chu trình.....	242
P_4_08_5.PAS * Thuật toán Floyd	245
P_4_09_1.PAS * Thuật toán Kruskal	253
P_4_09_2.PAS * Thuật toán Prim.....	257
P_4_10_1.PAS * Thuật toán Ford-Fulkerson.....	268
P_4_10_2.PAS * Thuật toán Preflow-push	273
P_4_11_1.PAS * Thuật toán đường mở tìm bộ ghép cực đại	286
P_4_12_1.PAS * Thuật toán Hungari.....	298
P_4_12_2.PAS * Cài đặt phương pháp Kuhn-Munkres $O(k^3)$	303
P_4_13_1.PAS * Phương pháp Lawler áp dụng cho thuật toán Edmonds	313

BẢNG CÁC KÝ HIỆU ĐƯỢC SỬ DỤNG

$\lfloor x \rfloor$	Floor of x: Số nguyên lớn nhất $\leq x$
$\lceil x \rceil$	Ceiling of x: Số nguyên nhỏ nhất $\geq x$
${}_n P_k$	Số chỉnh hợp không lặp chập k của n phần tử = $\frac{n!}{(n-k)!}$
$\binom{n}{k}$	Binomial coefficient: Hệ số của hạng tử x^k trong đa thức $(x+1)^n$ = Số tổ hợp chập k của n phần tử = $\frac{n!}{k!(n-k)!}$
$O(.)$	Ký pháp chữ O lớn
$\Theta(.)$	Ký pháp Θ lớn
$\Omega(.)$	Ký pháp Ω lớn
$o(.)$	Ký pháp chữ o nhỏ
$\omega(.)$	ký pháp ω nhỏ
$a[i..j]$	Các phần tử trong mảng a tính từ chỉ số i đến chỉ số j
$n!$	n factorial: Giai thừa của n = 1.2.3...n
$a \uparrow b$	a^b
$a \uparrow\uparrow b$	$\underbrace{a}_{b \text{ copies of } a}^{a \dots a}$
$\log_a x$	Logarithm to base a of x: Logarithm cơ số a của x ($\log_a a^b = b$)
$\lg x$	Logarithm nhị phân (cơ số 2) của x
$\ln x$	Logarithm tự nhiên (cơ số e) của x
$\log_a^* x$	Số lần lấy logarithm cơ số a để thu được số ≤ 1 từ x ($\log_a^*(a \uparrow\uparrow b) = b$)
$\lg^* x$	$\log_2^* x$
$\ln^* x$	$\log_e^* x$



PHẦN 1. BÀI TOÁN LIỆT KÊ

Có một số bài toán trên thực tế yêu cầu chỉ rõ: trong một tập các đối tượng cho trước có bao nhiêu đối tượng thoả mãn những điều kiện nhất định. Bài toán đó gọi là **bài toán đếm**.

Trong lớp các bài toán đếm, có những bài toán còn yêu cầu chỉ rõ những câu hình tìm được thoả mãn điều kiện đã cho là những câu hình nào. Bài toán yêu cầu đưa ra danh sách các câu hình có thể có gọi là **bài toán liệt kê**.

Để giải bài toán liệt kê, cần phải xác định được một **thuật toán** để có thể theo đó lần lượt xây dựng được tất cả các câu hình đang quan tâm. Có nhiều phương pháp liệt kê, nhưng chúng cần phải đáp ứng được hai yêu cầu dưới đây:

- Không được lặp lại một câu hình
- Không được bỏ sót một câu hình

Có thể nói rằng, phương pháp liệt kê là phương pháp cuối cùng để giải được một số bài toán tổ hợp hiện nay. Khó khăn chính của phương pháp này chính là sự bùng nổ tổ hợp dẫn tới sự đòi hỏi lớn về không gian và thời gian thực hiện chương trình. Tuy nhiên cùng với sự phát triển của máy tính điện tử, bằng phương pháp liệt kê, nhiều bài toán tổ hợp đã tìm thấy lời giải. Qua đó, ta cũng nên biết rằng **chỉ nên dùng phương pháp liệt kê khi không còn một phương pháp nào khác** tìm ra lời giải. Chính những nỗ lực giải quyết các bài toán thực tế không dùng phương pháp liệt kê đã thúc đẩy sự phát triển của nhiều ngành toán học.

§1. NHẮC LẠI MỘT SỐ KIẾN THỨC ĐẠI SỐ TỔ HỢP

Cho S là một tập hữu hạn gồm n phần tử và k là một số tự nhiên.

Gọi X là tập các số nguyên dương từ 1 đến k : $X = \{1, 2, \dots, k\}$

1.1. CHỈNH HỢP LẶP

Mỗi ánh xạ $f: X \rightarrow S$. Cho tương ứng với mỗi $i \in X$, một và chỉ một phần tử $f(i) \in S$.

Được gọi là một chỉnh hợp lặp chập k của S .

Nhưng do X là tập hữu hạn (k phần tử) nên ánh xạ f có thể xác định qua bảng các giá trị $f(1), f(2), \dots, f(k)$.

Ví dụ: $S = \{A, B, C, D, E, F\}; k = 3$. Một ánh xạ f có thể cho như sau:

i	1	2	3
$f(i)$	E	C	E

Vậy có thể đồng nhất f với dãy giá trị $(f(1), f(2), \dots, f(k))$ và coi dãy giá trị này cũng là một chỉnh hợp lặp chập k của S . Như ví dụ trên (E, C, E) là một chỉnh hợp lặp chập 3 của S . Để dàng chứng minh được kết quả sau bằng quy nạp hoặc bằng phương pháp đánh giá khả năng lựa chọn:

Số chỉnh hợp lặp chập k của tập gồm n phần tử là n^k

1.2. CHỈNH HỢP KHÔNG LẶP

Khi f là đơn ánh có nghĩa là với $\forall i, j \in X$ ta có $f(i) = f(j) \Leftrightarrow i = j$. Nói một cách dễ hiểu, khi dãy giá trị $f(1), f(2), \dots, f(k)$ gồm các phần tử thuộc S khác nhau đôi một thì f được gọi là một chỉnh hợp không lặp chập k của S . Ví dụ một chỉnh hợp không lặp (C, A, E) :

i	1	2	3
$f(i)$	C	A	E

Số chỉnh hợp không lặp chập k của tập gồm n phần tử là:

$${}_n P_k = n(n-1)(n-2)\dots(n-k+1) = \frac{n!}{(n-k)!}$$

1.3. HOÁN VỊ

Khi $k = n$. Một chỉnh hợp không lặp chập n của S được gọi là một hoán vị các phần tử của S .

Ví dụ: một hoán vị: $\langle A, D, C, E, B, F \rangle$ của $S = \{A, B, C, D, E, F\}$

i	1	2	3	4	5	6
$f(i)$	A	D	C	E	B	F

Để ý rằng khi $k = n$ thì số phần tử của tập $X = \{1, 2, \dots, n\}$ đúng bằng số phần tử của S . Do tính chất đôi một khác nhau nên dãy $f(1), f(2), \dots, f(n)$ sẽ liệt kê được hết các phần tử trong S . Như vậy f là toàn ánh. Mặt khác do giả thiết f là chỉnh hợp không lặp nên f là đơn ánh. Ta có tương ứng 1-1 giữa các phần tử của X và S , do đó f là song ánh. Vậy nên ta có thể định nghĩa một hoán vị của S là một song ánh giữa $\{1, 2, \dots, n\}$ và S .

Số hoán vị của tập gồm n phần tử = số chỉnh hợp không lặp chập $n = n!$

1.4. TỔ HỢP

Một tập con gồm k phần tử của S được gọi là một tổ hợp chập k của S .

Lấy một tập con k phần tử của S , xét tất cả $k!$ hoán vị của tập con này. Dễ thấy rằng các hoán vị đó là các chỉnh hợp không lặp chập k của S . Ví dụ lấy tập $\{A, B, C\}$ là tập con của tập S trong ví dụ trên thì: $\langle A, B, C \rangle, \langle C, A, B \rangle, \langle B, C, A \rangle, \dots$ là các chỉnh hợp không lặp chập 3 của S . Điều đó tức là khi liệt kê tất cả các chỉnh hợp không lặp chập k thì mỗi tổ hợp chập k sẽ được tính $k!$ lần. Vậy số tổ hợp chập k của tập gồm n phần tử là $\frac{n!}{k!(n-k)!} = \binom{n}{k}$

§2. PHƯƠNG PHÁP SINH (GENERATION)

Phương pháp sinh có thể áp dụng để giải bài toán liệt kê tổ hợp đặt ra nếu như hai điều kiện sau thoả mãn:

- ❖ Có thể xác định được một thứ tự trên tập các cấu hình tổ hợp cần liệt kê. Từ đó có thể biết được cấu hình đầu tiên và cấu hình cuối cùng trong thứ tự đó.
- ❖ Xây dựng được thuật toán từ một cấu hình chưa phải cấu hình cuối, sinh ra được cấu hình kế tiếp nó.

Phương pháp sinh có thể mô tả như sau:

```
(Xây dựng cấu hình đầu tiên);
repeat
    (Đưa ra cấu hình đang có);
    (Từ cấu hình đang có sinh ra cấu hình kế tiếp nếu còn);
until (hết cấu hình);
```

Thứ tự từ điển

Trên các kiểu dữ liệu đơn giản chuẩn, người ta thường nói tới khái niệm thứ tự. Ví dụ trên kiểu số thì có quan hệ: $1 < 2$; $2 < 3$; $3 < 10$; ..., trên kiểu ký tự Char thì cũng có quan hệ ' $A' < 'B'$ '; ' $C' < 'c'$ '...

Xét quan hệ thứ tự toàn phần “nhỏ hơn hoặc bằng” ký hiệu “ \leq ” trên một tập hợp S , là quan hệ hai ngôi thoả mãn bốn tính chất:

Với $\forall a, b, c \in S$

- ❖ Tính phô biến: Hoặc là $a \leq b$, hoặc $b \leq a$;
- ❖ Tính phản xạ: $a \leq a$
- ❖ Tính phản đối xứng: Nếu $a \leq b$ và $b \leq a$ thì bắt buộc $a = b$.
- ❖ Tính bắc cầu: Nếu có $a \leq b$ và $b \leq c$ thì $a \leq c$.

Trong trường hợp $a \leq b$ và $a \neq b$, ta dùng ký hiệu “ $<$ ” cho gọn, (ta ngầm hiểu các ký hiệu như \geq , $>$, khỏi phải định nghĩa)

Ví dụ như quan hệ “ \leq ” trên các số nguyên cũng như trên các kiểu vô hướng, liệt kê là quan hệ thứ tự toàn phần.

Trên các dãy hữu hạn, người ta cũng xác định một quan hệ thứ tự:

Xét $a[1..n]$ và $b[1..n]$ là hai dãy độ dài n , trên các phần tử của a và b đã có quan hệ thứ tự “ \leq ”.

Khi đó $a \leq b$ nếu như

Hoặc $a[i] = b[i]$ với $\forall i: 1 \leq i \leq n$.

Hoặc tồn tại một số nguyên dương $k: 1 \leq k < n$ để:

$$a[1] = b[1]$$

$$a[2] = b[2]$$

...

$$a[k-1] = b[k-1]$$

$$a[k] = b[k]$$

$$a[k+1] < b[k+1]$$

Trong trường hợp này, ta có thể viết $a < b$.

Thứ tự đó gọi là **thứ tự từ điển** trên các dãy độ dài n.

Khi độ dài hai dãy a và b không bằng nhau, người ta cũng xác định được thứ tự từ điển. Bằng cách thêm vào cuối dãy a hoặc dãy b những phần tử đặc biệt gọi là phần tử \emptyset để độ dài của a và b bằng nhau, và coi những phần tử \emptyset này nhỏ hơn tất cả các phần tử khác, ta lại đưa về xác định thứ tự từ điển của hai dãy cùng độ dài. Ví dụ:

$$\langle 1, 2, 3, 4 \rangle < \langle 5, 6 \rangle$$

$$\langle a, b, c \rangle < \langle a, b, c, d \rangle$$

$$'calculator' < 'computer'$$

2.1. SINH CÁC DÃY NHỊ PHÂN ĐỘ DÀI N

Một dãy nhị phân độ dài n là một dãy $x[1..n]$ trong đó $x[i] \in \{0, 1\}$ ($\forall i : 1 \leq i \leq n$).

Dễ thấy: một dãy nhị phân x độ dài n là biểu diễn nhị phân của một giá trị nguyên $p(x)$ nào đó nằm trong đoạn $[0, 2^n - 1]$. Số các dãy nhị phân độ dài n = số các số tự nhiên $\in [0, 2^n - 1] = 2^n$. Ta sẽ lập chương trình liệt kê các dãy nhị phân theo thứ tự từ điển có nghĩa là sẽ liệt kê lần lượt các dãy nhị phân biểu diễn các số nguyên theo thứ tự 0, 1, ..., $2^n - 1$.

Ví dụ: Khi $n = 3$, các dãy nhị phân độ dài 3 được liệt kê như sau:

$p(x)$	0	1	2	3	4	5	6	7
x	000	001	010	011	100	101	110	111

Như vậy dãy đầu tiên sẽ là 00...0 và dãy cuối cùng sẽ là 11...1. Nhận xét rằng nếu dãy x = $x[1..n]$ là dãy đang có và không phải dãy cuối cùng cần liệt kê thì dãy kế tiếp sẽ nhận được bằng cách cộng thêm 1 (theo cơ số 2 có nhớ) vào dãy hiện tại.

Ví dụ khi $n = 8$:

$$\text{Dãy đang có: } 10010000$$

$$\text{cộng thêm 1: } \quad + 1$$

$$\text{Dãy đang có: } 10010111$$

$$\text{cộng thêm 1: } \quad + 1$$

$$\text{Dãy mới: } 10010001$$

$$\text{Dãy mới: } 10011000$$

Như vậy kỹ thuật sinh cấu hình kế tiếp từ cấu hình hiện tại có thể mô tả như sau: Xét từ cuối dãy về đầu (xét từ hàng đơn vị lên), tìm số 0 gấp đầu tiên

- ❖ Nếu thấy thì thay số 0 đó bằng số 1 và đặt tất cả các phần tử phía sau vị trí đó bằng 0.
- ❖ Nếu không thấy thì toàn dãy là số 1, đây là cấu hình cuối cùng

Dữ liệu vào (**Input**): nhập từ file văn bản BSTR.INP chứa số nguyên dương $n \leq 100$

Kết quả ra (**Output**): ghi ra file văn bản BSTR.OUT các dãy nhị phân độ dài n .

BSTR. INP	BSTR. OUT
3	000
	001
	010
	011
	100
	101
	110
	111

P_1_02_1.PAS * Thuật toán sinh liệt kê các dãy nhị phân độ dài n

```
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Binary.Strings;
const
  InputFile = 'BSTR.INP';
  OutputFile = 'BSTR.OUT';
  max = 100;
var
  x: array[1..max] of Integer;
  n, i: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n);
  Close(f);
  Assign(f, OutputFile); Rewrite(f);
  FillChar(x, SizeOf(x), 0); {Cấu hình ban đầu x=00..0}
  repeat {Thuật toán sinh}
    for i := 1 to n do Write(f, x[i]); {In ra cấu hình hiện tại}
    WriteLn(f);
    i := n; {x[i] là phần tử cuối dãy, lùi dần i cho tới khi gặp số 0 hoặc khi i = 0 thì dừng}
    while (i > 0) and (x[i] = 1) do Dec(i);
    if i > 0 then {Chưa gặp phải cấu hình 11...1}
      begin
        x[i] := 1; {Thay x[i] bằng số 1}
        FillChar(x[i + 1], (n - i) * SizeOf(x[1]), 0); {Đặt x[i+1] = x[i+2] = ... = x[n] := 0}
      end;
    until i = 0; {Đã hết cấu hình}
  Close(f);
end.
```

2.2. LIỆT KÊ CÁC TẬP CON K PHẦN TỬ

Ta sẽ lập chương trình liệt kê các tập con k phần tử của tập $\{1, 2, \dots, n\}$ theo thứ tự từ điển

Ví dụ: với $n = 5$, $k = 3$, ta phải liệt kê đủ 10 tập con:

1. {1, 2, 3} 2. {1, 2, 4} 3. {1, 2, 5} 4. {1, 3, 4} 5. {1, 3, 5}
6. {1, 4, 5} 7. {2, 3, 4} 8. {2, 3, 5} 9. {2, 4, 5} 10. {3, 4, 5}

Như vậy tập con đầu tiên (cấu hình khởi tạo) là $\{1, 2, \dots, k\}$.

Cấu hình kết thúc là $\{n - k + 1, n - k + 2, \dots, n\}$.

Nhận xét: Ta sẽ in ra tập con bằng cách in ra lần lượt các phần tử của nó theo thứ tự tăng dần. Biểu diễn mỗi tập con là một dãy $x[1..k]$ trong đó $x[1] < x[2] < \dots < x[k]$. Ta nhận thấy giới

hạn trên (giá trị lớn nhất có thể nhận) của $x[k]$ là n , của $x[k-1]$ là $n - 1$, của $x[k-2]$ là $n - 2 \dots$
 Tổng quát: giới hạn trên của $x[i] = n - k + i$;

Còn tất nhiên, giới hạn dưới của $x[i]$ (giá trị nhỏ nhất $x[i]$ có thể nhận) là $x[i-1] + 1$.

Như vậy nếu ta đang có một dãy x đại diện cho một tập con, nếu x là cấu hình kết thúc có nghĩa là tất cả các phần tử trong x đều đã đạt tới giới hạn trên thì quá trình sinh kết thúc, nếu không thì ta phải sinh ra một dãy x mới tăng dần thoả mãn vừa đủ lớn hơn dãy cũ theo nghĩa không có một tập con k phần tử nào chen giữa chúng khi sắp thứ tự từ điển.

Ví dụ: $n = 9$, $k = 6$. Cấu hình đang có $x = \langle 1, 2, 6, 7, 8, 9 \rangle$. Các phần tử $x[3]$ đến $x[6]$ đã đạt tới giới hạn nên để sinh cấu hình mới ta không thể sinh bằng cách tăng một phần tử trong số các $x[6], x[5], x[4], x[3]$ lên được, ta phải tăng $x[2] = 2$ lên thành $x[2] = 3$. Được cấu hình mới là $x = \langle 1, 3, 6, 7, 8, 9 \rangle$. Cấu hình này đã thoả mãn lớn hơn cấu hình trước nhưng chưa thoả mãn tính chất vừa đủ lớn muốn vậy ta lại thay $x[3], x[4], x[5], x[6]$ bằng các giới hạn dưới của nó. Tức là:

$$x[3] := x[2] + 1 = 4$$

$$x[4] := x[3] + 1 = 5$$

$$x[5] := x[4] + 1 = 6$$

$$x[6] := x[5] + 1 = 7$$

Ta được cấu hình mới $x = \langle 1, 3, 4, 5, 6, 7 \rangle$ là cấu hình kế tiếp. Nếu muốn tìm tiếp, ta lại nhận thấy rằng $x[6] = 7$ chưa đạt giới hạn trên, như vậy chỉ cần tăng $x[6]$ lên 1 là được $x = \langle 1, 3, 4, 5, 6, 8 \rangle$.

Vậy kỹ thuật sinh tập con kế tiếp từ tập đã có x có thể xây dựng như sau:

Tìm từ cuối dãy lên đầu cho tới khi gặp một phần tử $x[i]$ chưa đạt giới hạn trên $n - k + i$.

❖ Nếu tìm thấy:

Tăng $x[i]$ đó lên 1.

Đặt tất cả các phần tử phía sau $x[i]$ bằng giới hạn dưới.

❖ Nếu không tìm thấy tức là mọi phần tử đã đạt giới hạn trên, đây là cấu hình cuối cùng

Input: file văn bản SUBSET.INP chứa hai số nguyên dương n, k ($1 \leq k \leq n \leq 100$) cách nhau ít nhất một dấu cách

Output: file văn bản SUBSET.OUT các tập con k phần tử của tập $\{1, 2, \dots, n\}$

SUBSET.INP	SUBSET.OUT
5 3	$\{1, 2, 3\}$
	$\{1, 2, 4\}$
	$\{1, 2, 5\}$
	$\{1, 3, 4\}$
	$\{1, 3, 5\}$
	$\{1, 4, 5\}$
	$\{2, 3, 4\}$
	$\{2, 3, 5\}$
	$\{2, 4, 5\}$
	$\{3, 4, 5\}$

```

P_1_02_2.PAS * Thuật toán sinh liệt kê các tập con k phần tử
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Combination;
const
  InputFile = 'SUBSET.INP';
  OutputFile = 'SUBSET.OUT';
  max = 100;
var
  x: array[1..max] of Integer;
  n, k, i, j: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n, k);
  Close(f);
  Assign(f, OutputFile); Rewrite(f);
  for i := 1 to k do x[i] := i; {Khởi tạo x := (1, 2, ..., k)}
  repeat
    {In ra cấu hình hiện tại}
    Write(f, '{');
    for i := 1 to k - 1 do Write(f, x[i], ', ');
    WriteLn(f, x[k], '}');
    {Sinh tiếp}
    i := k; {Xét từ cuối dãy lên tìm x[i] chưa đạt giới hạn trên n - k + i}
    while (i > 0) and (x[i] = n - k + i) do Dec(i);
    if i > 0 then {Nếu chưa lùi đến 0 có nghĩa là chưa phải cấu hình kết thúc}
      begin
        Inc(x[i]); {Tăng x[i] lên 1, Đặt các phần tử đứng sau x[i] bằng giới hạn dưới của nó}
        for j := i + 1 to k do x[j] := x[j - 1] + 1;
      end;
    until i = 0; {Lùi đến tận 0 có nghĩa là tất cả các phần tử đã đạt giới hạn trên - hết cấu hình}
  Close(f);
end.

```

2.3. LIỆT KÊ CÁC HOÁN VỊ

Ta sẽ lập chương trình liệt kê các hoán vị của $\{1, 2, \dots, n\}$ theo thứ tự từ điển.

Ví dụ với $n = 4$, ta phải liệt kê đủ 24 hoán vị:

```

1.1234 2.1243 3.1324 4.1342 5.1423 6.1432
7.2134 8.2143 9.2314 10.2341 11.2413 12.2431
13.3124 14.3142 15.3214 16.3241 17.3412 18.3421
19.4123 20.4132 21.4213 22.4231 23.4312 24.4321

```

Như vậy hoán vị đầu tiên sẽ là $\langle 1, 2, \dots, n \rangle$. Hoán vị cuối cùng là $\langle n, n-1, \dots, 1 \rangle$.

Hoán vị sẽ sinh ra phải lớn hơn hoán vị hiện tại, hơn thế nữa phải là hoán vị vừa đủ lớn hơn hoán vị hiện tại theo nghĩa không thể có một hoán vị nào khác chen giữa chúng khi sắp thứ tự.

Giả sử hoán vị hiện tại là $x = \langle 3, 2, 6, 5, 4, 1 \rangle$, xét 4 phần tử cuối cùng, ta thấy chúng được xếp giảm dần, điều đó có nghĩa là cho dù ta có hoán vị 4 phần tử này thế nào, ta cũng được một hoán vị bé hơn hoán vị hiện tại. Như vậy ta phải xét đến $x[2] = 2$, thay nó bằng một giá trị khác. Ta sẽ thay bằng giá trị nào?, không thể là 1 bởi nếu vậy sẽ được hoán vị nhỏ hơn, không thể là 3 vì đã có $x[1] = 3$ rồi (phần tử sau không được chọn vào những giá trị mà phần tử trước đã chọn). Còn lại các giá trị 4, 5, 6. Vì cần một hoán vị vừa đủ lớn hơn hiện tại nên ta chọn $x[2] = 4$. Còn các giá trị ($x[3], x[4], x[5], x[6]$) sẽ lấy trong tập $\{2, 6, 5, 1\}$. Cũng vì tính vừa

đủ lớn nên ta sẽ tìm biểu diễn nhỏ nhất của 4 số này gán cho $x[3], x[4], x[5], x[6]$ tức là $\langle 1, 2, 5, 6 \rangle$. Vậy hoán vị mới sẽ là $\langle 3, 4, 1, 2, 5, 6 \rangle$.

Ta có nhận xét gì qua ví dụ này: Đoạn cuối của hoán vị hiện tại được xếp giảm dần, số $x[5] = 4$ là số nhỏ nhất trong đoạn cuối giảm dần thoả mãn điều kiện lớn hơn $x[2] = 2$. Nếu đổi chỗ $x[5]$ cho $x[2]$ thì ta sẽ được $x[2] = 4$ và đoạn cuối vẫn được sắp xếp giảm dần. Khi đó muốn biểu diễn nhỏ nhất cho các giá trị trong đoạn cuối thì ta chỉ cần đảo ngược đoạn cuối.

Trong trường hợp hoán vị hiện tại là $\langle 2, 1, 3, 4 \rangle$ thì hoán vị kế tiếp sẽ là $\langle 2, 1, 4, 3 \rangle$. Ta cũng có thể coi hoán vị $\langle 2, 1, 3, 4 \rangle$ có đoạn cuối giảm dần, đoạn cuối này chỉ gồm 1 phần tử (4)

Vậy kỹ thuật sinh hoán vị kế tiếp từ hoán vị hiện tại có thể xây dựng như sau:

Xác định đoạn cuối giảm dần dài nhất, tìm chỉ số i của phần tử $x[i]$ đứng liền trước đoạn cuối đó. Điều này đồng nghĩa với việc tìm từ vị trí sát cuối dãy lên đầu, gấp chỉ số i đầu tiên thỏa mãn $x[i] < x[i+1]$.

❖ Nếu tìm thấy chỉ số i như trên

Trong đoạn cuối giảm dần, tìm phần tử $x[k]$ nhỏ nhất thoả mãn điều kiện $x[k] > x[i]$. Do đoạn cuối giảm dần, điều này thực hiện bằng cách tìm từ cuối dãy lên đầu gấp chỉ số k đầu tiên thoả mãn $x[k] > x[i]$ (có thể dùng tìm kiếm nhị phân).

Đảo giá trị $x[k]$ và $x[i]$

Lật ngược thứ tự đoạn cuối giảm dần (từ $x[i+1]$ đến $x[k]$) trở thành tăng dần.

❖ Nếu không tìm thấy tức là toàn dãy đã sắp giảm dần, đây là cấu hình cuối cùng

Input: file văn bản PERMUTE.INP chứa số nguyên dương $n \leq 100$

Output: file văn bản PERMUTE.OUT các hoán vị của dãy $(1, 2, \dots, n)$

PERMUTE.INP	PERMUTE.OUT
3	1 2 3
	1 3 2
	2 1 3
	2 3 1
	3 1 2
	3 2 1

P_1_02_3.PAS * Thuật toán sinh liệt kê hoán vị

```
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Permutation;
const
  InputFile = 'PERMUTE.INP';
  OutputFile = 'PERMUTE.OUT';
  max = 100;
var
  n, i, k, a, b: Integer;
  x: array[1..max] of Integer;
  f: Text;

procedure Swap(var X, Y: Integer); {Thủ tục đảo giá trị hai tham biến X, Y}
var
  Temp: Integer;
begin
  Temp := X; X := Y; Y := Temp;
end;
```

```

begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n);
  Close(f);
  Assign(f, OutputFile); Rewrite(f);
  for i := 1 to n do x[i] := i; {Khởi tạo cấu hình đầu: x[1] := 1; x[2] := 2; ..., x[n] := n}
repeat
  for i := 1 to n do Write(f, x[i], ' '); {In ra cấu hình hoán vị hiện tại}
  WriteLn(f);
  i := n - 1;
  while (i > 0) and (x[i] > x[i + 1]) do Dec(i);
  if i > 0 then {Chưa gặp phải hoán vị cuối (n, n-1, ..., 1)}
    begin
      k := n; {x[k] là phần tử cuối dãy}
      while x[k] < x[i] do Dec(k); {Lùi dần k để tìm gặp x[k] đầu tiên lớn hơn x[i]}
      Swap(x[k], x[i]); {Đổi chỗ x[k] và x[i]}
      a := i + 1; b := n; {Lật ngược đoạn cuối giảm dần, a: đầu đoạn, b: cuối đoạn}
      while a < b do
        begin
          Swap(x[a], x[b]); {Đảo giá trị x[a] và x[b]}
          Inc(a); {Tiến a và lùi b, tiếp tục cho tới khi a, b chạm nhau}
          Dec(b);
        end;
      end;
    until i = 0; {Toàn dãy là dãy giảm dần - không sinh tiếp được - hết cấu hình}
  Close(f);
end.

```

Bài tập:**Bài 1**

Các chương trình trên xử lý không tốt trong trường hợp tầm thường, đó là trường hợp $n = 0$ đối với chương trình liệt kê dãy nhị phân cũng như trong chương trình liệt kê hoán vị, trường hợp $k = 0$ đối với chương trình liệt kê tổ hợp, hãy khắc phục điều đó.

Bài 2

Liệt kê các dãy nhị phân độ dài n có thể coi là liệt kê các chỉnh hợp lặp chập n của tập 2 phần tử $\{0, 1\}$. Hãy lập chương trình:

Nhập vào hai số n và k , liệt kê các chỉnh hợp lặp chập k của $\{0, 1, \dots, n-1\}$.

Hướng dẫn: thay hệ cơ số 2 bằng hệ cơ số n .

Bài 3

Hãy liệt kê các dãy nhị phân độ dài n mà trong đó cụm chữ số “01” xuất hiện đúng 2 lần.

Bài 4.

Nhập vào một danh sách n tên người. Liệt kê tất cả các cách chọn ra đúng k người trong số n người đó.

Bài 5

Liệt kê tất cả các tập con của tập $\{1, 2, \dots, n\}$. Có thể dùng phương pháp liệt kê tập con như trên hoặc dùng phương pháp liệt kê tất cả các dãy nhị phân. Mỗi số 1 trong dãy nhị phân tương ứng với một phần tử được chọn trong tập. Ví dụ với tập $\{1, 2, 3, 4\}$ thì dãy nhị phân

1010 sẽ tương ứng với tập con $\{1, 3\}$. Hãy lập chương trình in ra tất cả các tập con của $\{1, 2, \dots, n\}$ theo hai phương pháp.

Bài 6

Nhập vào danh sách tên n người, in ra tất cả các cách xếp n người đó vào một bàn

Bài 7

Nhập vào danh sách n bạn nam và n bạn nữ, in ra tất cả các cách xếp $2n$ người đó vào một bàn tròn, mỗi bạn nam tiếp đến một bạn nữ.

Bài 8

Người ta có thể dùng phương pháp sinh để liệt kê các chỉnh hợp không lặp chập k. Tuy nhiên có một cách khác là liệt kê tất cả các tập con k phần tử của tập hợp, sau đó in ra đủ k! hoán vị của nó. Hãy viết chương trình liệt kê các chỉnh hợp không lặp chập k của $\{1, 2, \dots, n\}$ theo cả hai cách.

Bài 9

Liệt kê tất cả các hoán vị chữ cái trong từ MISSISSIPPI theo thứ tự từ điển.

Bài 10

Liệt kê tất cả các cách phân tích số nguyên dương n thành tổng các số nguyên dương, hai cách phân tích là hoán vị của nhau chỉ tính là một cách.

Cuối cùng, ta có nhận xét, mỗi phương pháp liệt kê đều có ưu, nhược điểm riêng và phương pháp sinh cũng không nằm ngoài nhận xét đó. Phương pháp sinh không thể sinh ra được cấu hình thứ p nếu như chưa có cấu hình thứ p - 1, chứng tỏ rằng phương pháp sinh tỏ ra ưu điểm trong trường hợp liệt kê toàn bộ một số lượng nhỏ cấu hình trong một bộ dữ liệu lớn thì lại có nhược điểm và ít tính phổ dụng trong những thuật toán duyệt hạn chế. Hơn thế nữa, không phải cấu hình ban đầu lúc nào cũng dễ tìm được, không phải kỹ thuật sinh cấu hình kế tiếp cho mọi bài toán đều đơn giản như trên (Sinh các chỉnh hợp không lặp chập k theo thứ tự từ điển chẳng hạn). Ta sang một chuyên mục sau nói đến một phương pháp liệt kê có tính phổ dụng cao hơn, để giải các bài toán liệt kê phức tạp hơn đó là: Thuật toán quay lui (Back tracking).

§3. THUẬT TOÁN QUAY LUI

Thuật toán quay lui dùng để giải bài toán liệt kê các cấu hình. Mỗi cấu hình được xây dựng bằng cách xây dựng từng phần tử, mỗi phần tử được chọn bằng cách thử tất cả các khả năng. Giả sử cấu hình cần liệt kê có dạng $x[1..n]$, khi đó thuật toán quay lui thực hiện qua các bước:

- 1) Xét tất cả các giá trị $x[1]$ có thể nhận, thử cho $x[1]$ nhận lần lượt các giá trị đó. Với mỗi giá trị thử gán cho $x[1]$ ta sẽ:
 - 2) Xét tất cả các giá trị $x[2]$ có thể nhận, lại thử cho $x[2]$ nhận lần lượt các giá trị đó. Với mỗi giá trị thử gán cho $x[2]$ lại xét tiếp các khả năng chọn $x[3] \dots$ cứ tiếp tục như vậy đến bước:
- ...
- n) Xét tất cả các giá trị $x[n]$ có thể nhận, thử cho $x[n]$ nhận lần lượt các giá trị đó, thông báo cấu hình tìm được $\langle x[1], x[2], \dots, x[n] \rangle$.

Trên phương diện quy nạp, có thể nói rằng thuật toán quay lui liệt kê các cấu hình n phần tử dạng $x[1..n]$ bằng cách thử cho $x[1]$ nhận lần lượt các giá trị có thể. Với mỗi giá trị thử gán cho $x[1]$ bài toán trở thành liệt kê tiếp cấu hình n - 1 phần tử $x[2..n]$.

Mô hình của thuật toán quay lui có thể mô tả như sau:

{Thử tục này thử cho $x[i]$ nhận lần lượt các giá trị mà nó có thể nhận}

```
procedure Attempt(i);
begin
  for {mỗi giá trị V có thể gán cho  $x[i]$ } do
    begin
      {Thử cho  $x[i] := V$ };
      if { $x[i]$  là phần tử cuối cùng trong cấu hình} then
        {Thông báo cấu hình tìm được}
      else
        begin
          {Ghi nhận việc cho  $x[i]$  nhận giá trị V (nếu cần)};
          Attempt(i + 1); {Gọi đệ quy để chọn tiếp  $x[i+1]$ }
          {Nếu cần, bỏ ghi nhận việc thử  $x[i] := V$  để thử giá trị khác};
        end;
    end;
end;
```

Thuật toán quay lui sẽ bắt đầu bằng lời gọi `Attempt(1)`

3.1. LIỆT KÊ CÁC DÃY NHỊ PHÂN ĐỘ DÀI N

Input/Output với khuôn dạng như trong P_1_02_1.PAS

Biểu diễn dãy nhị phân độ dài N dưới dạng $x[1..n]$. Ta sẽ liệt kê các dãy này bằng cách thử dùng các giá trị {0, 1} gán cho $x[i]$. Với mỗi giá trị thử gán cho $x[i]$ lại thử các giá trị có thể gán cho $x[i+1]$. Chương trình liệt kê bằng thuật toán quay lui có thể viết:

```
P_1_03_1.PAS * Thuật toán quay lui liệt kê các dãy nhị phân độ dài n
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program BinaryStrings;
const
```

```


InputFile = 'BSTR.INP';
OutputFile = 'BSTR.OUT';
max = 100;
var
  x: array[1..max] of Integer;
  n: Integer;
  f: Text;

procedure PrintResult; {In câu hình tìm được, do thủ tục tìm đệ quy Attempt gọi khi tìm ra một câu hình}
var
  i: Integer;
begin
  for i := 1 to n do Write(f, x[i]);
  WriteLn(f);
end;

procedure Attempt(i: Integer); {Thử các cách chọn x[i]}
var
  j: Integer;
begin
  for j := 0 to 1 do {Xét các giá trị có thể gán cho x[i], với mỗi giá trị đó}
    begin
      x[i] := j; {Thử đặt x[i]}
      if i = n then PrintResult {Nếu i = n thì in kết quả}
      else Attempt(i + 1); {Nếu i chưa phải là phần tử cuối thì tìm tiếp x[i+1]}
    end;
end;
end;
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n); {Nhập dữ liệu}
  Close(f);
  Assign(f, OutputFile); Rewrite(f);
  Attempt(1); {Thử các cách chọn giá trị x[1]}
  Close(f);
end.

```

Ví dụ: Khi $n = 3$, cây tìm kiếm quay lui như sau:



Hình 1: Cây tìm kiếm quay lui trong bài toán liệt kê dãy nhị phân

3.2. LIỆT KÊ CÁC TẬP CON K PHẦN TỬ

Input/Output có khuôn dạng như trong P_1_02_2.PAS

Để liệt kê các tập con k phần tử của tập $S = \{1, 2, \dots, n\}$ ta có thể dựa vào liệt kê các cấu hình $x[1..n]$, ở đây các $x[i] \in S$ và $x[1] < x[2] < \dots < x[k]$. Ta có nhận xét:

$$x[k] \leq n$$

$$x[k-1] \leq x[k] - 1 \leq n - 1$$

...

$$x[i] \leq n - k + i$$

...

$$x[1] \leq n - k + 1.$$

Từ đó suy ra $x[i-1] + 1 \leq x[i] \leq n - k + i$ ($1 \leq i \leq k$) ở đây ta giả thiết có thêm một số $x[0] = 0$ khi xét $i = 1$.

Như vậy ta sẽ xét tất cả các cách chọn $x[1]$ từ 1 ($=x[0] + 1$) đến $n - k + 1$, với mỗi giá trị đó, xét tiếp tất cả các cách chọn $x[2]$ từ $x[1] + 1$ đến $n - k + 2$, ... cứ như vậy khi chọn được đến $x[k]$ thì ta có một cấu hình cần liệt kê. Chương trình liệt kê bằng thuật toán quay lui như sau:

```
P_1_03_2.PAS * Thuật toán quay lui liệt kê các tập con k phần tử
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Combination;
const
  InputFile = 'SUBSET.INP';
  OutputFile = 'SUBSET.OUT';
  max = 100;
var
  x: array[0..max] of Integer;
  n, k: Integer;
  f: Text;

procedure PrintResult; (*In ra tập con {x[1], x[2], ..., x[k]}*)
var
  i: Integer;
begin
  Write(f, '{');
  for i := 1 to k - 1 do Write(f, x[i], ', ');
  WriteLn(f, x[k], '}');
end;

procedure Attempt(i: Integer); {Thử các cách chọn giá trị cho x[i]}
var
  j: Integer;
begin
  for j := x[i - 1] + 1 to n - k + i do
    begin
      x[i] := j;
      if i = k then PrintResult
      else Attempt(i + 1);
    end;
end;

begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n, k);
  Close(f);
  Assign(f, OutputFile); Rewrite(f);
```

```

x[0] := 0;
Attempt(1);
Close(f);
end.

```

Nếu để ý chương trình trên và chương trình liệt kê dãy nhị phân độ dài n, ta thấy về cơ bản chúng chỉ khác nhau ở thủ tục Attempt(i) - chọn thử các giá trị cho $x[i]$, ở chương trình liệt kê dãy nhị phân ta thử chọn các giá trị 0 hoặc 1 còn ở chương trình liệt kê các tập con k phần tử ta thử chọn $x[i]$ là một trong các giá trị nguyên từ $x[i-1] + 1$ đến $n - k + i$. Qua đó ta có thể thấy tính phổ dụng của thuật toán quay lui: mô hình cài đặt có thể thích hợp cho nhiều bài toán, khác với phương pháp sinh tuần tự, với mỗi bài toán lại phải có một thuật toán sinh kế tiếp riêng làm cho việc cài đặt mỗi bài một khác, bên cạnh đó, không phải thuật toán sinh kế tiếp nào cũng dễ cài đặt.

3.3. LIỆT KÊ CÁC CHỈNH HỢP KHÔNG LẶP CHẬP K

Để liệt kê các chỉnh hợp không lặp chập k của tập $S = \{1, 2, \dots, n\}$ ta có thể đưa về liệt kê các cấu hình $x[1..k]$ ở đây các $x[i] \in S$ và khác nhau đôi một.

Như vậy thủ tục Attempt(i) - xét tất cả các khả năng chọn $x[i]$ - sẽ thử hết các giá trị từ 1 đến n, mà các giá trị này chưa bị các phần tử đứng trước chọn. Muốn xem các giá trị nào chưa được chọn ta sử dụng kỹ thuật dùng mảng đánh dấu:

- ❖ Khởi tạo một mảng $c[1..n]$ mang kiểu logic boolean. Ở đây $c[i]$ cho biết giá trị i có còn tự do hay đã bị chọn rồi. Ban đầu khởi tạo tất cả các phần tử mảng c là TRUE có nghĩa là các phần tử từ 1 đến n đều tự do.
- ❖ Tại bước chọn các giá trị có thể của $x[i]$ ta chỉ xét những giá trị j có $c[j] = \text{TRUE}$ có nghĩa là chỉ chọn những giá trị tự do.
- ❖ Trước khi gọi đệ quy tìm $x[i+1]$: ta đặt giá trị j vừa gán cho $x[i]$ là **đã bị chọn** có nghĩa là đặt $c[j] := \text{FALSE}$ để các thủ tục Attempt($i + 1$), Attempt($i + 2$)... gọi sau này không chọn phải giá trị j đó nữa
- ❖ Sau khi gọi đệ quy tìm $x[i+1]$: có nghĩa là sắp tới ta sẽ thử gán một **giá trị khác** cho $x[i]$ thì ta sẽ đặt giá trị j vừa thử đó thành **tự do** ($c[j] := \text{TRUE}$), bởi khi xi đã nhận một giá trị khác rồi thì các phần tử đứng sau: $x[i+1], x[i+2], \dots$ hoàn toàn có thể nhận lại giá trị j đó. Điều này hoàn toàn hợp lý trong phép xây dựng chỉnh hợp không lặp: $x[1]$ có n cách chọn, $x[2]$ có $n - 1$ cách chọn, ... Lưu ý rằng khi thủ tục Attempt(i) có $i = k$ thì ta không cần phải đánh dấu gì cả vì tiếp theo chỉ có in kết quả chứ không cần phải chọn thêm phần tử nào nữa.

Input: file văn bản ARRANGE.INP chứa hai số nguyên dương n, k ($1 \leq k \leq n \leq 100$) cách nhau ít nhất một dấu cách

Output: file văn bản ARRANGE.OUT ghi các chỉnh hợp không lặp chập k của tập $\{1, \dots, n\}$

ARRANGE.INP	ARRANGE.OUT
3 2	1 2
	1 3
	2 1
	2 3
	3 1
	3 2

P_1_03_3.PAS * Thuật toán quay lui liệt kê các chinh hợp không lặp chap k

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Arrangement;
const
  InputFile = 'ARRANGES.INP';
  OutputFile = 'ARRANGES.OUT';
  max = 100;
var
  x: array[1..max] of Integer;
  c: array[1..max] of Boolean;
  n, k: Integer;
  f: Text;

procedure PrintResult; {Thù tục in câu hình tìm được}
var
  i: Integer;
begin
  for i := 1 to k do Write(f, x[i], ' ');
  WriteLn(f);
end;

procedure Attempt(i: Integer); {Thù các cách chọn x[i]}
var
  j: Integer;
begin
  for j := 1 to n do
    if c[j] then {Chi xét những giá trị j còn tự do}
      begin
        x[i] := j;
        if i = k then PrintResult {Nếu đã chọn được đến xk thì chỉ việc in kết quả}
        else
          begin
            c[j] := False; {Đánh dấu: j đã bị chọn}
            Attempt(i + 1); {Thù tục này chỉ xét những giá trị còn tự do gán cho x[i+1]}
            c[j] := True; {Bỏ đánh dấu: j lại là tự do, bởi sắp tới sẽ thử một cách chọn khác của x[i]}
          end;
      end;
end;
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n, k);
  Assign(f, OutputFile); Rewrite(f);
  FillChar(c, SizeOf(c), True); {Tất cả các số đều chưa bị chọn}
  Attempt(1); {Thù các cách chọn giá trị của x[1]}
  Close(f);
end.

```

Nhận xét: khi k = n thì đây là chương trình liệt kê hoán vị

3.4. BÀI TOÁN PHÂN TÍCH SỐ

3.4.1. Bài toán

Cho một số nguyên dương $n \leq 30$, hãy tìm tất cả các cách phân tích số n thành tổng của các số nguyên dương, các cách phân tích là hoán vị của nhau chỉ tính là 1 cách.

3.4.2. Cách làm:

Ta sẽ lưu nghiệm trong mảng x , ngoài ra có một mảng t . Mảng t xây dựng như sau: $t[i]$ sẽ là tổng các phần tử trong mảng x từ $x[1]$ đến $x[i]$: $t[i] := x[1] + x[2] + \dots + x[i]$.

Khi liệt kê các dãy x có tổng các phần tử đúng bằng n , để tránh sự trùng lặp ta đưa thêm ràng buộc $x[i-1] \leq x[i]$.

Vì số phần tử thực sự của mảng x là không cố định nên thủ tục `PrintResult` dùng để in ra 1 cách phân tích phải có thêm tham số cho biết sẽ in ra bao nhiêu phần tử.

Thủ tục đệ quy `Attempt(i)` sẽ thử các giá trị có thể nhận của $x[i]$ ($x[i] \geq x[i - 1]$)

Khi nào thì in kết quả và khi nào thì gọi đệ quy tìm tiếp ?

Lưu ý rằng $t[i - 1]$ là tổng của tất cả các phần tử từ $x[1]$ đến $x[i-1]$ do đó

- ❖ Khi $t[i] = n$ tức là ($x[i] = n - t[i - 1]$) thì in kết quả
- ❖ Khi tìm tiếp, $x[i+1]$ sẽ phải lớn hơn hoặc bằng $x[i]$. Mặt khác $t[i+1]$ là tổng của các số từ $x[1]$ tới $x[i+1]$ không được vượt quá n . Vậy ta có $t[i+1] \leq n \Leftrightarrow t[i-1] + x[i] + x[i+1] \leq n \Leftrightarrow x[i] + x[i+1] \leq n - t[i-1]$ tức là $x[i] \leq (n - t[i-1])/2$. Ví dụ đơn giản khi $n = 10$ thì chọn $x[1] = 6, 7, 8, 9$ là việc làm vô nghĩa vì như vậy cũng không ra nghiệm mà cũng không chọn tiếp $x[2]$ được nữa.

Một cách dễ hiểu: ta gọi đệ quy tìm tiếp khi giá trị $x[i]$ được chọn còn cho phép chọn thêm một phần tử khác lớn hơn hoặc bằng nó mà không làm tổng vượt quá n . Còn ta in kết quả chỉ khi $x[i]$ mang giá trị đúng bằng số thiểu hụt của tổng i-1 phần tử đầu so với n .

Vậy thủ tục `Attempt(i)` thử các giá trị cho $x[i]$ có thể viết như sau: (để tổng quát cho $i = 1$, ta đặt $x[0] = 1$ và $t[0] = 0$).

- ❖ Xét các giá trị của $x[i]$ từ $x[i - 1]$ đến $(n - t[i-1]) \text{ div } 2$, cập nhật $t[i] := t[i - 1] + x[i]$ và gọi đệ quy tìm tiếp.
- ❖ Cuối cùng xét giá trị $x[i] = n - t[i-1]$ và in kết quả từ $x[1]$ đến $x[i]$.

Input: file văn bản ANALYSE.INP chứa số nguyên dương $n \leq 100$

Output: file văn bản ANALYSE.OUT ghi các cách phân tích số n .

ANALYSE.INP	ANALYSE.OUT
6	6 = 1+1+1+1+1+1
	6 = 1+1+1+1+2
	6 = 1+1+1+3
	6 = 1+1+2+2
	6 = 1+1+4
	6 = 1+2+3
	6 = 1+5
	6 = 2+2+2
	6 = 2+4
	6 = 3+3
	6 = 6

P_1_03_4.PAS * Thuật toán quay lui liệt kê các cách phân tích số

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Analyses;
const
  InputFile = 'ANALYSE.INP';
  OutputFile = 'ANALYSE.OUT';
  max = 100;
var
  n: Integer;
  x: array[0..max] of Integer;
  t: array[0..max] of Integer;
  f: Text;

procedure Init; {Khởi tạo}
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n);
  Close(f);
  x[0] := 1;
  t[0] := 0;
end;

procedure PrintResult(k: Integer);
var
  i: Integer;
begin
  Write(f, n, ' = ');
  for i := 1 to k - 1 do Write(f, x[i], '+');
  WriteLn(f, x[k]);
end;

procedure Attempt(i: Integer);
var
  j: Integer;
begin
  for j := x[i - 1] to (n - T[i - 1]) div 2 do {Trường hợp còn chọn tiếp x[i+1]}
    begin
      x[i] := j;
      t[i] := t[i - 1] + j;
      Attempt(i + 1);
    end;
  x[i] := n - T[i - 1]; {Nếu x[i] là phần tử cuối thì nó bắt buộc phải là n-T[i-1], in kết quả}
  PrintResult(i);
end;

begin
  Init;
  Assign(f, OutputFile); Rewrite(f);
  Attempt(1);
  Close(f);
end;

```

end.


Bây giờ ta xét tiếp một ví dụ kinh điển của thuật toán quay lui:

3.5. BÀI TOÁN XẾP HẬU

3.5.1. Bài toán

Xét bàn cờ tổng quát kích thước nxn. Một quân hậu trên bàn cờ có thể ăn được các quân khác nằm tại các ô cùng hàng, cùng cột hoặc cùng đường chéo. Hãy tìm các xếp n quân hậu trên bàn cờ sao cho không quân nào ăn quân nào.

Ví dụ một cách xếp với n = 8:




Hình 2: Xếp 8 quân hậu trên bàn cờ 8x8

3.5.2. Phân tích

Rõ ràng n quân hậu sẽ được đặt mỗi con một hàng vì hậu ăn được ngang, ta gọi quân hậu sẽ đặt ở hàng 1 là quân hậu 1, quân hậu ở hàng 2 là quân hậu 2... quân hậu ở hàng n là quân hậu n. Vậy một nghiệm của bài toán sẽ được biết khi ta tìm ra được **vị trí cột của những quân hậu**.

Nếu ta định hướng Đông (Phải), Tây (Trái), Nam (Dưới), Bắc (Trên) thì ta nhận thấy rằng:

- ❖ Một đường chéo theo hướng Đông Bắc - Tây Nam (ĐB-TN) bất kỳ sẽ đi qua một số ô, các ô đó có tính chất: Hàng + Cột = C (Const). Với mỗi đường chéo ĐB-TN ta có 1 hàng số C và với một hàng số C: $2 \leq C \leq 2n$ xác định duy nhất 1 đường chéo ĐB-TN vì vậy ta có thể đánh chỉ số cho các đường chéo ĐB-TN từ 2 đến $2n$
- ❖ Một đường chéo theo hướng Đông Nam - Tây Bắc (ĐN-TB) bất kỳ sẽ đi qua một số ô, các ô đó có tính chất: Hàng - Cột = C (Const). Với mỗi đường chéo ĐN-TB ta có 1 hàng số C và với một hàng số C: $1 - n \leq C \leq n - 1$ xác định duy nhất 1 đường chéo ĐN-TB vì vậy ta có thể đánh chỉ số cho các đường chéo ĐN-TB từ $1 - n$ đến $n - 1$.



Hình 3: Đường chéo ĐB-TN mang chỉ số 10 và đường chéo ĐN-TB mang chỉ số 0

Cài đặt:

Ta có 3 mảng logic để đánh dấu:

- ❖ Mảng $a[1..n]$. $a[i] = \text{TRUE}$ nếu như cột i còn tự do, $a[i] = \text{FALSE}$ nếu như cột i đã bị một quân hậu không chế
- ❖ Mảng $b[2..2n]$. $b[i] = \text{TRUE}$ nếu như đường chéo ĐB-TN thứ i còn tự do, $b[i] = \text{FALSE}$ nếu như đường chéo đó đã bị một quân hậu không chế.
- ❖ Mảng $c[1..n-1]$. $c[i] = \text{TRUE}$ nếu như đường chéo ĐN-TB thứ i còn tự do, $c[i] = \text{FALSE}$ nếu như đường chéo đó đã bị một quân hậu không chế.

Ban đầu cả 3 mảng đánh dấu đều mang giá trị TRUE. (Các cột và đường chéo đều tự do)

Thuật toán quay lui:

- ❖ Xét tất cả các cột, thử đặt quân hậu 1 vào một cột, với mỗi cách đặt như vậy, xét tất cả các cách đặt quân hậu 2 không bị quân hậu 1 ăn, lại thử 1 cách đặt và xét tiếp các cách đặt quân hậu 3... Mỗi cách đặt được đến quân hậu n cho ta 1 nghiệm
- ❖ Khi chọn vị trí cột j cho quân hậu thứ i , thì ta phải chọn ô (i, j) không bị các quân hậu đặt trước đó ăn, tức là phải chọn cột j còn tự do, đường chéo ĐB-TN $(i+j)$ còn tự do, đường chéo ĐN-TB $(i-j)$ còn tự do. Điều này có thể kiểm tra ($a[j] = b[i+j] = c[i-j] = \text{TRUE}$)
- ❖ Khi thử đặt được quân hậu thứ i vào cột j , nếu đó là quân hậu cuối cùng ($i = n$) thì ta có một nghiệm. Nếu không:

Trước khi gọi đệ quy tìm cách đặt quân hậu thứ $i + 1$, ta đánh dấu cột và 2 đường chéo bị quân hậu vừa đặt không chế ($a[j] = b[i+j] = c[i-j] := \text{FALSE}$) để các lần gọi đệ quy tiếp sau chọn cách đặt các quân hậu kế tiếp sẽ không chọn vào những ô nằm trên cột j và những đường chéo này nữa.

Sau khi gọi đệ quy tìm cách đặt quân hậu thứ $i + 1$, có nghĩa là sắp tới ta lại thử một cách đặt khác cho quân hậu thứ i , ta bỏ đánh dấu cột và 2 đường chéo bị quân hậu vừa thử đặt không chế ($a[j] = b[i+j] = c[i-j] := \text{TRUE}$) tức là cột và 2 đường chéo đó lại

thành tự do, bởi khi đã đặt quân hậu i sang vị trí khác rồi thì cột và 2 đường chéo đó hoàn toàn có thể gán cho một quân hậu khác

Hãy xem lại trong các chương trình liệt kê chỉnh hợp không lặp và hoán vị về kỹ thuật đánh dấu. Ở đây chỉ khác với liệt kê hoán vị là: liệt kê hoán vị chỉ cần một mảng đánh dấu xem giá trị có tự do không, còn bài toán xếp hậu thì cần phải đánh dấu cả 3 thành phần: Cột, đường chéo ĐB-TN, đường chéo ĐN- TB. Trường hợp đơn giản hơn: Yêu cầu liệt kê các cách đặt n quân xe lên bàn cờ nxn sao cho không quân nào ăn quân nào chính là bài toán liệt kê hoán vị

❖ **Input:** file văn bản QUEENS.INP chứa số nguyên dương $n \leq 100$

❖ **Output:** file văn bản QUEENS.OUT, mỗi dòng ghi một cách đặt n quân hậu

QUEENS.INP	QUEENS.OUT
5	(1, 1); (2, 3); (3, 5); (4, 2); (5, 4); (1, 1); (2, 4); (3, 2); (4, 5); (5, 3); (1, 2); (2, 4); (3, 1); (4, 3); (5, 5); (1, 2); (2, 5); (3, 3); (4, 1); (5, 4); (1, 3); (2, 1); (3, 4); (4, 2); (5, 5); (1, 3); (2, 5); (3, 2); (4, 4); (5, 1); (1, 4); (2, 1); (3, 3); (4, 5); (5, 2); (1, 4); (2, 2); (3, 5); (4, 3); (5, 1); (1, 5); (2, 2); (3, 4); (4, 1); (5, 3); (1, 5); (2, 3); (3, 1); (4, 4); (5, 2);

P_1_03_5.PAS * Thuật toán quay lui giải bài toán xếp hậu

```
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program n_Queens;
const
  InputFile = 'QUEENS.INP';
  OutputFile = 'QUEENS.OUT';
  max = 100;
var
  n: Integer;
  x: array[1..max] of Integer;
  a: array[1..max] of Boolean;
  b: array[2..2 * max] of Boolean;
  c: array[1 - max..max - 1] of Boolean;
  f: Text;

procedure Init;
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n);
  Close(f);
  FillChar(a, SizeOf(a), True); {Mọi cột đều tự do}
  FillChar(b, SizeOf(b), True); {Mọi đường chéo Đông Bắc - Tây Nam đều tự do}
  FillChar(c, SizeOf(c), True); {Mọi đường chéo Đông Nam - Tây Bắc đều tự do}
end;

procedure PrintResult;
var
  i: Integer;
begin
  for i := 1 to n do Write(f, '(', i, ', ', x[i], ')');
  WriteLn(f);
end;

procedure Attempt(i: Integer); {Thử các cách đặt quân hậu thứ i vào hàng i}
var
  j: Integer;
```

```

begin
  for j := 1 to n do
    if a[j] and b[i + j] and c[i - j] then {Chi xét những cột j mà ô (i, j) chưa bị khống chế}
      begin
        x[i] := j; {Thử đặt quân hậu i vào cột j}
        if i = n then PrintResult
        else
          begin
            a[j] := False; b[i + j] := False; c[i - j] := False; {Đánh dấu}
            Attempt(i + 1); {Tim các cách đặt quân hậu thứ i + 1}
            a[j] := True; b[i + j] := True; c[i - j] := True; {Bỏ đánh dấu}
          end;
      end;
  end;

begin
  Init;
  Assign(f, OutputFile); Rewrite(f);
  Attempt(1);
  Close(f);
end.

```

Tên gọi thuật toán quay lui, đứng trên phương diện cài đặt có thể nên gọi là kỹ thuật vét cạn bằng quay lui thì chính xác hơn, tuy nhiên đứng trên phương diện bài toán, nếu như ta coi công việc giải bài toán bằng cách xét tất cả các khả năng cũng là 1 cách giải thì tên gọi Thuật toán quay lui cũng không có gì trái logic. Xét hoạt động của chương trình trên cây tìm kiếm quay lui ta thấy tại bước thử chọn $x[i]$ nó sẽ gọi đệ quy để tìm tiếp $x[i+1]$ có nghĩa là quá trình sẽ duyệt tiến sâu xuống phía dưới đến tận nút lá, sau khi đã duyệt hết các nhánh, tiến trình lùi lại thử áp đặt một giá trị khác cho $x[i]$, đó chính là nguồn gốc của tên gọi “thuật toán quay lui”

Bài tập:

Bài 1

Một số chương trình trên xử lý không tốt trong trường hợp tầm thường ($n = 0$ hoặc $k = 0$), hãy khắc phục các lỗi đó

Bài 2

Viết chương trình liệt kê các chỉnh hợp lặp chập k của n phần tử

Bài 3

Cho hai số nguyên dương l, n . Hãy liệt kê các xâu nhị phân độ dài n có tính chất, bất kỳ hai xâu con nào độ dài l liền nhau đều khác nhau.

Bài 4

Với $n = 5$, $k = 3$, vẽ cây tìm kiếm quay lui của chương trình liệt kê tổ hợp chập k của tập $\{1, 2, \dots, n\}$

Bài 5

Liệt kê tất cả các tập con của tập S gồm n số nguyên $\{S[1], S[2], \dots, S[n]\}$ nhập vào từ bàn phím

Bài 6

Tương tự như bài 5 nhưng chỉ liệt kê các tập con có $\max - \min \leq T$ (T cho trước).

Bài 7

Một dãy $x[1..n]$ gọi là một hoán vị hoàn toàn của tập $\{1, 2, \dots, n\}$ nếu nó là một hoán vị và thoả mãn $x[i] \neq i$ với $\forall i: 1 \leq i \leq n$. Hãy viết chương trình liệt kê tất cả các hoán vị hoàn toàn của tập trên (n vào từ bàn phím).

Bài 8

Sửa lại thủ tục in kết quả (PrintResult) trong bài xếp hậu để có thể vẽ hình bàn cờ và các cách đặt hậu ra màn hình.

Bài 9

Mã đi tuần: Cho bàn cờ tổng quát kích thước $n \times n$ và một quân Mã, hãy chỉ ra một hành trình của quân Mã xuất phát từ ô đang đứng đi qua tất cả các ô còn lại của bàn cờ, mỗi ô đúng 1 lần.

Bài 10

Chuyển tất cả các bài tập trong bài trước đang viết bằng sinh tuần tự sang quay lui.

Bài 11

Xét sơ đồ giao thông gồm n nút giao thông đánh số từ 1 tới n và m đoạn đường nối chúng, mỗi đoạn đường nối 2 nút giao thông. Hãy nhập dữ liệu về mạng lưới giao thông đó, nhập số hiệu hai nút giao thông s và d . Hãy in ra tất cả các cách đi từ s tới d mà mỗi cách đi không được qua nút giao thông nào quá một lần.

§4. KỸ THUẬT NHÁNH CẬN

4.1. BÀI TOÁN TỐI ƯU

Một trong những bài toán đặt ra trong thực tế là việc tìm ra **một** nghiệm thoả mãn một số điều kiện nào đó, và nghiệm đó là **tốt nhất** theo một chỉ tiêu cụ thể, nghiên cứu lời giải các lớp bài toán tối ưu thuộc về lĩnh vực quy hoạch toán học. Tuy nhiên cũng cần phải nói rằng trong nhiều trường hợp chúng ta chưa thể xây dựng một thuật toán nào thực sự hữu hiệu để giải bài toán, mà cho tới nay việc tìm nghiệm của chúng vẫn phải dựa trên mô hình **liệt kê** toàn bộ các cấu hình có thể và đánh giá, tìm ra cấu hình tốt nhất. Việc liệt kê cấu hình có thể cài đặt bằng các phương pháp liệt kê: Sinh tuần tự và tìm kiếm quay lui. Dưới đây ta sẽ tìm hiểu phương pháp liệt kê bằng thuật toán quay lui để tìm nghiệm của bài toán tối ưu.

4.2. SỰ BÙNG NỔ TỒ HỢP

Mô hình thuật toán quay lui là tìm kiếm trên 1 cây phân cấp. Nếu giả thiết rằng ứng với mỗi nút tương ứng với một giá trị được chọn cho $x[i]$ sẽ ứng với chỉ 2 nút tương ứng với 2 giá trị mà $x[i+1]$ có thể nhận thì cây n cấp sẽ có tới 2^n nút lá, con số này lớn hơn rất nhiều lần so với dữ liệu đầu vào n . Chính vì vậy mà nếu như ta có thao tác thừa trong việc chọn $x[i]$ thì sẽ phải trả giá rất lớn về chi phí thực thi thuật toán bởi quá trình tìm kiếm lòng vòng vô nghĩa trong các bước chọn kế tiếp $x[i+1], x[i+2], \dots$. Khi đó, một vấn đề đặt ra là trong quá trình liệt kê lời giải ta cần tận dụng những thông tin đã tìm được để loại bỏ sớm những phương án chắc chắn không phải tối ưu. Kỹ thuật đó gọi là kỹ thuật đánh giá nhánh cận trong tiến trình quay lui.

4.3. MÔ HÌNH KỸ THUẬT NHÁNH CẬN

Dựa trên mô hình thuật toán quay lui, ta xây dựng mô hình sau:

```

procedure Init;
begin
  {Khởi tạo một cấu hình bất kỳ BESTCONFIG};
end;

{Thủ tục này thử chọn cho x[i] tất cả các giá trị nó có thể nhận}
procedure Attempt(i: Integer);
begin
  for {Mỗi giá trị V có thể gán cho x[i]} do
    begin
      {Thử cho x[i] := V};
      if {Việc thử trên vẫn còn hi vọng tìm ra cấu hình tốt hơn BESTCONFIG} then
        if {x[i] là phần tử cuối cùng trong cấu hình} then
          {Cập nhật BESTCONFIG}
        else
          begin
            {Ghi nhận việc thử x[i] = V nếu cần};
            Attempt(i + 1); {Gọi đệ quy, chọn tiếp x[i+1]}
            {Bỏ ghi nhận việc thử cho x[i] = V (nếu cần)};
          end;
    end;
end;

```

```

begin
  Init;
  Attempt(1);
  {Thông báo cấu hình tối ưu BESTCONFIG};
end.

```

Kỹ thuật nhánh cặn thêm vào cho thuật toán quay lui khả năng đánh giá theo từng bước, nếu tại bước thứ i, giá trị thử gán cho $x[i]$ không có hi vọng tìm thấy cấu hình tốt hơn cấu hình BESTCONFIG thì thử giá trị khác ngay mà không cần phải gọi đệ quy tìm tiếp hay ghi nhận kết quả làm gì. Nghiệm của bài toán sẽ được làm tốt dần, bởi khi tìm ra một cấu hình mới (tốt hơn BESTCONFIG - tất nhiên), ta không in kết quả ngay mà sẽ cập nhật BESTCONFIG bằng cấu hình mới vừa tìm được

4.4. BÀI TOÁN NGƯỜI DU LỊCH

4.4.1. Bài toán

Cho n thành phố đánh số từ 1 đến n và m tuyến đường giao thông hai chiều giữa chúng, mạng lưới giao thông này được cho bởi bảng C cấp $n \times n$, ở đây $C[i, j] = C[j, i] =$ Chi phí đi đoạn đường trực tiếp từ thành phố i đến thành phố j . Giả thiết rằng $C[i, i] = 0$ với $\forall i$, $C[i, j] = +\infty$ nếu không có đường trực tiếp từ thành phố i đến thành phố j .

Một người du lịch xuất phát từ thành phố 1, muốn đi thăm tất cả các thành phố còn lại mỗi thành phố đúng 1 lần và cuối cùng quay lại thành phố 1. Hãy chỉ ra cho người đó hành trình với chi phí ít nhất. Bài toán đó gọi là bài toán người du lịch hay bài toán hành trình của một thương gia (Traveling Salesman)

4.4.2. Cách giải

Hành trình cần tìm có dạng $x[1..n + 1]$ trong đó $x[1] = x[n + 1] = 1$ ở đây giữa $x[i]$ và $x[i+1]$: hai thành phố liên tiếp trong hành trình phải có đường đi trực tiếp ($C[i, j] \neq +\infty$) và ngoại trừ thành phố 1, không thành phố nào được lặp lại hai lần. Có nghĩa là dãy $x[1..n]$ lập thành 1 hoán vị của $(1, 2, \dots, n)$.

Duyệt quay lui: $x[2]$ có thể chọn một trong các thành phố mà $x[1]$ có đường đi tới (trực tiếp), với mỗi cách thử chọn $x[2]$ như vậy thì $x[3]$ có thể chọn một trong các thành phố mà $x[2]$ có đường đi tới (ngoài $x[1]$). Tổng quát: $x[i]$ có thể chọn 1 trong các thành phố **chưa đi qua** mà từ $x[i-1]$ có đường đi trực tiếp tới ($1 \leq i \leq n$).

Nhánh cặn: Khi tạo cấu hình BestConfig có chi phí = $+\infty$. Với mỗi bước thử chọn $x[i]$ xem chi phí đường đi cho tới lúc đó có < Chi phí của cấu hình BestConfig?, nếu không nhỏ hơn thì thử giá trị khác ngay bởi có đi tiếp cũng chỉ tốn thêm. Khi thử được một giá trị $x[n]$ ta kiểm tra xem $x[n]$ có đường đi trực tiếp về 1 không? Nếu có đánh giá chi phí đi từ thành phố 1 đến thành phố $x[n]$ cộng với chi phí từ $x[n]$ đi trực tiếp về 1, nếu nhỏ hơn chi phí của đường đi BestConfig thì cập nhật lại BestConfig bằng cách đi mới.


Sau thủ tục tìm kiếm quay lui mà chi phí của BestConfig vẫn bằng $+\infty$ thì có nghĩa là nó không tìm thấy một hành trình nào thoả mãn điều kiện để bài để cập nhật BestConfig, bài toán

không có lời giải, còn nếu chi phí của BestConfig < $+\infty$ thì in ra cấu hình BestConfig - đó là hành trình ít tốn kém nhất tìm được

Input: file văn bản TOURISM.INP

- ❖ Dòng 1: Chứa số thành phố n ($1 \leq n \leq 100$) và số tuyến đường m trong mạng lưới giao thông
- ❖ m dòng tiếp theo, mỗi dòng ghi số hiệu hai thành phố có đường đi trực tiếp và chi phí đi trên quãng đường đó (chi phí này là số nguyên dương ≤ 10000)

Output: file văn bản TOURISM.OUT, ghi hành trình tìm được.



P_1_04_1.PAS * Kỹ thuật nhánh cận dùng cho bài toán người du lịch

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program TravellingSalesman;
const
  InputFile = 'TOURISM.INP';
  OutputFile = 'TOURISM.OUT';
  max = 100;
  maxE = 10000;
  maxC = max * maxE;{+∞}
var
  C: array[1..max, 1..max] of Integer; {Ma trận chi phí}
  X, BestWay: array[1..max + 1] of Integer; {X để thù các khả năng, BestWay để ghi nhận nghiệm}
  T: array[1..max + 1] of Integer; {T[i] để lưu chi phí đi từ X[1] đến X[i]}
  Free: array[1..max] of Boolean; {Free để đánh dấu, Free[i]= True nếu chưa đi qua tp i}
  m, n: Integer;
  MinSpending: Integer; {Chi phí hành trình tối ưu}

procedure Enter;
var
  i, j, k: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n, m);
  for i := 1 to n do {Khởi tạo bảng chi phí ban đầu}
    for j := 1 to n do
      if i = j then C[i, j] := 0 else C[i, j] := maxC;
  for k := 1 to m do
    begin
      ReadLn(f, i, j, C[i, j]);
      C[j, i] := C[i, j]; {Chi phí như nhau trên 2 chiều}
    end;
  Close(f);
end;

procedure Init; {Khởi tạo}
begin
  FillChar(Free, n, True);
  Free[1] := False; {Các thành phố là chưa đi qua ngoại trừ thành phố 1}
end;

```

```

X[1] := 1; {Xuất phát từ thành phố 1}
T[1] := 0; {Chi phí tại thành phố xuất phát là 0}
MinSpending := maxC;
end;

procedure Attempt(i: Integer); {Thử các cách chọn x[i]}
var
  j: Integer;
begin
  for j := 2 to n do {Thử các thành phố từ 2 đến n}
    if Free[j] then {Nếu gặp thành phố chưa đi qua}
      begin
        X[i] := j; {Thứ đi}
        T[i] := T[i - 1] + C[x[i - 1], j]; {Chi phí := Chi phí bước trước + chi phí đường đi trực tiếp}
        if T[i] < MinSpending then {Hiện nhiên nếu có điều này thì C[x[i - 1], j] < +∞ rồi}
          if i < n then {Nếu chưa đến được x[n]}
            begin
              Free[j] := False; {Đánh dấu thành phố vừa thử}
              Attempt(i + 1); {Tim các khả năng chọn x[i+1]}
              Free[j] := True; {Bỏ đánh dấu}
            end
          else
            if T[n] + C[x[n], 1] < MinSpending then {Từ x[n] quay lại 1 vẫn tồn chi phí ít hơn trước}
              begin {Cập nhật BestConfig}
                BestWay := X;
                MinSpending := T[n] + C[x[n], 1];
              end;
            end;
          end;
        end;
      end;
    end;

procedure PrintResult;
var
  i: Integer;
  f: Text;
begin
  Assign(f, OutputFile); Rewrite(f);
  if MinSpending = maxC then WriteLn(f, 'NO SOLUTION')
  else
    for i := 1 to n do Write(f, BestWay[i], '->');
  WriteLn(f, 1);
  WriteLn(f, 'Cost: ', MinSpending);
  Close(f);
end;

begin
  Enter;
  Init;
  Attempt(2);
  PrintResult;
end.

```

Trên đây là một giải pháp nhánh cận còn rất thô sơ giải bài toán người du lịch, trên thực tế người ta còn có nhiều cách đánh giá nhánh cận chặt hơn nữa. Hãy tham khảo các tài liệu khác để tìm hiểu về những phương pháp đó.

4.5. DÃY ABC

Cho trước một số nguyên dương N ($N \leq 100$), hãy tìm một xâu chỉ gồm các ký tự A, B, C thoả mãn 3 điều kiện:

- ❖ Có độ dài N

- ❖ Hai đoạn con bất kỳ liền nhau đều khác nhau (đoạn con là một dãy ký tự liên tiếp của xâu)
- ❖ Có ít ký tự C nhất.

Cách giải:

Không trình bày, đề nghị tự xem chương trình để hiểu, chỉ chú thích kỹ thuật nhánh cận như sau:

Nếu dãy X[1..n] thoả mãn 2 đoạn con bất kỳ liền nhau đều khác nhau, thì trong 4 ký tự liên tiếp bất kỳ bao giờ cũng phải có 1 ký tự “C”. Như vậy với một dãy con gồm k ký tự liên tiếp của dãy X thì số ký tự C trong dãy con đó bắt buộc phải $\geq k \text{ div } 4$.

Tại bước thử chọn X[i], nếu ta đã có T[i] ký tự “C” trong đoạn đã chọn từ X[1] đến X[i], thì cho dù các bước đệ quy tiếp sau làm tốt như thế nào chăng nữa, số ký tự “C” sẽ phải chọn thêm bao giờ cũng $\geq (n - i) \text{ div } 4$. Tức là nếu theo phương án chọn X[i] như thế này thì số ký tự “C” trong dãy kết quả (khi chọn đến X[n]) cho dù có làm tốt đến đâu cũng $\geq T[i] + (n - i) \text{ div } 4$. Ta dùng con số này để đánh giá nhánh cận, nếu nó nhiều hơn số ký tự “C” trong BestConfig thì chắc chắn có làm tiếp cũng chỉ được một cấu hình tồi tệ hơn, ta bỏ qua ngay cách chọn này và thử phương án khác.

Input: file văn bản ABC.INP chứa số nguyên dương $n \leq 100$

Output: file văn bản ABC.OUT ghi xâu tìm được

ABC.INP	ABC.OUT
10	ABACABCBAB "C" Letter Count : 2

P_1_04_2.PAS * Dãy ABC

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program ABC_STRING;
const
  InputFile = 'ABC.INP';
  OutputFile = 'ABC.OUT';
  max = 100;
var
  N, MinC: Integer;
  X, Best: array[1..max] of 'A'..'C';
  T: array[0..max] of Integer; {T[i] cho biết số ký tự "C" trong đoạn từ X[1] đến X[i]}
  f: Text;

{Hàm Same(i, l) cho biết xâu gồm l ký tự kết thúc tại X[i] có trùng với xâu l ký tự liền trước nó không ?}
function Same(i, l: Integer): Boolean;
var
  j, k: Integer;
begin
  j := i - l; {j là vị trí cuối đoạn liền trước đoạn đó}
  for k := 0 to l - 1 do
    if X[i - k] <> X[j - k] then
      begin
        Same := False; Exit;
      end;
  Same := True;
end;

{Hàm Check(i) cho biết X[i] có làm hỏng tính không lặp của dãy X[1..i] hay không}
function Check(i: Integer): Boolean;
var

```

```

    l: Integer;
begin
  for l := 1 to i div 2 do {Thứ các độ dài l}
    if Same(i, l) then {Nếu có xâu độ dài l kết thúc bởi X[i] bị trùng với xâu liền trước}
      begin
        Check := False; Exit;
      end;
  Check := True;
end;

{Giữ lại kết quả vừa tìm được vào BestConfig (MinC và mảng Best)}
procedure KeepResult;
begin
  MinC := T[N];
  Best := X;
end;

{Thuật toán quay lui có nhánh cận}
procedure Attempt(i: Integer); {Thứ các giá trị có thể của X[i]}
var
  j: 'A'..'C';
begin
  for j := 'A' to 'C' do {Xét tất cả các giá trị}
    begin
      X[i] := j;
      if Check(i) then {Nếu thêm giá trị đó vào không làm hỏng tính không lặp}
        begin
          if j = 'C' then T[i] := T[i - 1] + 1 {Tính T[i] qua T[i - 1]}
          else T[i] := T[i - 1];
          if T[i] + (N - i) div 4 < MinC then {Đánh giá nhánh cận}
            if i = N then KeepResult
            else Attempt(i + 1);
        end;
    end;
end;

procedure PrintResult;
var
  i: Integer;
begin
  for i := 1 to N do Write(f, Best[i]);
  WriteLn(f);
  WriteLn(f, '"C" Letter Count : ', MinC);
end;

begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, N);
  Close(f);
  Assign(f, OutputFile); Rewrite(f);
  T[0] := 0;
  MinC := N; {Khởi tạo cấu hình BestConfig ban đầu rất tồi}
  Attempt(1);
  PrintResult;
  Close(f);
end.

```

Nếu ta thay bài toán là tìm xâu ít ký tự 'B' nhất mà vẫn viết chương trình tương tự như trên thì chương trình sẽ chạy chậm hơn chút ít. Lý do: thủ tục Attempt ở trên sẽ thử lần lượt các giá trị 'A', 'B', rồi mới đến 'C'. Có nghĩa ngay trong cách tìm, nó đã tiết kiệm sử dụng ký tự 'C' nhất nên trong phần lớn các bộ dữ liệu nó nhanh chóng tìm ra lời giải hơn so với bài toán tương

ứng tìm xâu ít ký tự 'B' nhất. Chính vì vậy mà nếu như đè bài yêu cầu ít ký tự 'B' nhất ta cứ lập chương trình làm yêu cầu ít ký tự 'C' nhất, chỉ có điều khi in kết quả, ta đổi vai trò 'B', 'C' cho nhau. Đây là một ví dụ cho thấy sức mạnh của thuật toán quay lui khi kết hợp với kỹ thuật nhánh cận, nếu viết quay lui thuần tuý hoặc đánh giá nhánh cận không tốt thì với $N = 100$, tôi cũng không đủ kiên nhẫn để đợi chương trình cho kết quả (chỉ biết rằng > 3 giờ). Trong khi đó khi $N = 100$, với chương trình trên chỉ chạy hết hơn 1 giây cho kết quả là xâu 27 ký tự 'C'.

Nói chung, ít khi ta gặp bài toán mà chỉ cần sử dụng một thuật toán, một mô hình kỹ thuật cài đặt là có thể giải được. Thông thường các bài toán thực tế đòi hỏi phải có sự tổng hợp, pha trộn nhiều thuật toán, nhiều kỹ thuật mới có được một lời giải tốt. Không được lạm dụng một kỹ thuật nào và cũng không xem thường một phương pháp nào khi bắt tay vào giải một bài toán tin học. Thuật toán quay lui cũng không phải là ngoại lệ, ta phải biết phối hợp một cách uyển chuyển với các thuật toán khác thì khi đó nó mới thực sự là một công cụ mạnh.

Bài tập:

Bài 1

Một dãy dấu ngoặc hợp lệ là một dãy các ký tự “(” và “)” được định nghĩa như sau:

- i. Dãy rỗng là một dãy dấu ngoặc hợp lệ độ sâu 0
- ii. Nếu A là dãy dấu ngoặc hợp lệ độ sâu k thì (A) là dãy dấu ngoặc hợp lệ độ sâu k + 1
- iii. Nếu A và B là hai dãy dấu ngoặc hợp lệ với độ sâu lần lượt là p và q thì AB là dãy dấu ngoặc hợp lệ độ sâu là $\max(p, q)$

Độ dài của một dãy ngoặc là tổng số ký tự “(” và “)”

Ví dụ: Có 5 dãy dấu ngoặc hợp lệ độ dài 8 và độ sâu 3:

1. (((()())
2. ((())())
3. ((())())()
4. ()((())
5. ()(((())

Bài toán đặt ra là khi cho biết trước hai số nguyên dương n và k. Hãy liệt kê hết các dãy ngoặc hợp lệ có độ dài là n và độ sâu là k (làm được với n càng lớn càng tốt).

Bài 2

Cho một bãi mìn kích thước mxn ô vuông, trên một ô có thể có chứa một quả mìn hoặc không, để biểu diễn bản đồ mìn đó, người ta có hai cách:

Cách 1: dùng bản đồ đánh dấu: sử dụng một lưới ô vuông kích thước mxn, trên đó tại ô (i, j) ghi số 1 nếu ô đó có mìn, ghi số 0 nếu ô đó không có mìn

Cách 2: dùng bản đồ mật độ: sử dụng một lưới ô vuông kích thước mxn, trên đó tại ô (i, j) ghi một số trong khoảng từ 0 đến 8 cho biết tổng số mìn trong các ô lân cận với ô (i, j) ((i, j) lân cận với ô (i, j) là ô có chung với ô (i, j) ít nhất 1 đỉnh).

Giả thiết rằng hai bản đồ được ghi chính xác theo tình trạng mìn trên hiện trường.

Về nguyên tắc, lúc cài bãi mìn phải vẽ cả bản đồ đánh dấu và bản đồ mật độ, tuy nhiên sau một thời gian dài, khi người ta muôn gỡ mìn ra khỏi bãi thì vẫn để hết sức khó khăn bởi bản đồ đánh dấu đã bị thất lạc !!. Công việc của các lập trình viên là: Từ bản đồ mật độ, hãy tái tạo lại bản đồ đánh dấu của bãi mìn.

Dữ liệu: Vào từ file văn bản MINE.INP, các số trên 1 dòng cách nhau ít nhất 1 dấu cách

- ❖ Dòng 1: Ghi 2 số nguyên dương m, n ($2 \leq m, n \leq 30$)
- ❖ m dòng tiếp theo, dòng thứ i ghi n số trên hàng i của bản đồ mật độ theo đúng thứ tự từ trái qua phải.

Kết quả: Ghi ra file văn bản MINE.OUT, các số trên 1 dòng ghi cách nhau ít nhất 1 dấu cách

- ❖ Dòng 1: Ghi tổng số lượng mìn trong bãi
- ❖ m dòng tiếp theo, dòng thứ i ghi n số trên hàng i của bản đồ đánh dấu theo đúng thứ tự từ trái qua phải.

Ví dụ:

MINE.INP	MINE.OUT
10 15	80
0 3 2 3 3 3 5 3 4 4 5 4 4 4 3	1 0 1 1 1 1 0 1 1 1 1 1 1 1 1
1 4 3 5 5 4 5 4 7 7 7 5 6 6 5	0 0 1 0 0 1 1 1 0 1 1 1 0 1 1
1 4 3 5 4 3 5 4 4 4 4 3 4 5 5	0 0 1 0 0 1 0 0 1 1 1 0 0 1 1
1 4 2 4 4 5 4 2 4 4 3 2 3 5 4	1 0 1 1 1 0 0 1 0 0 0 0 0 1 1
1 3 2 5 4 4 2 2 3 2 3 3 2 5 2	1 0 0 0 1 1 1 0 0 1 0 0 1 0 1
2 3 2 3 3 5 3 2 4 4 3 4 2 4 1	0 0 0 0 1 0 0 0 0 1 1 0 1 0 0
2 3 2 4 3 3 2 3 4 6 6 5 3 3 1	0 1 1 0 0 1 0 0 1 1 0 0 1 0 0
2 6 4 5 2 4 1 3 3 5 5 5 6 4 3	1 0 1 0 1 0 1 0 1 1 1 1 0 1 0
4 6 5 7 3 5 3 5 5 6 5 4 4 4 3	0 1 1 0 1 0 0 0 0 0 1 1 1 1 1
2 4 4 4 2 3 1 2 2 2 3 3 3 4 2	1 1 1 1 1 0 1 1 1 1 0 0 0 0 1



PHẦN 2. CẤU TRÚC DỮ LIỆU VÀ GIẢI THUẬT

Hạt nhân của các chương trình máy tính là sự lưu trữ và xử lý thông tin. Việc tổ chức dữ liệu như thế nào có ảnh hưởng rất lớn đến cách thức xử lý dữ liệu đó cũng như tốc độ thực thi và sự chiếm dụng bộ nhớ của chương trình. Việc đặc tả bằng các cấu trúc tổng quát (generic structures) và các kiểu dữ liệu trừu tượng (abstract data types) còn cho phép người lập trình có thể dễ dàng hình dung ra các công việc cụ thể và giảm bớt công sức trong việc chỉnh sửa, nâng cấp và sử dụng lại các thiết kế đã có. Mục đích của phần này là cung cấp những hiểu biết nền tảng trong việc thiết kế một chương trình máy tính, để thấy rõ được sự cần thiết của việc phân tích, lựa chọn cấu trúc dữ liệu phù hợp cho từng bài toán cụ thể; đồng thời khảo sát một số cấu trúc dữ liệu và thuật toán kinh điển mà lập trình viên nào cũng cần phải nắm vững.

§1. CÁC BƯỚC CƠ BẢN KHI TIẾN HÀNH GIẢI CÁC BÀI TOÁN TIN HỌC

1.1. XÁC ĐỊNH BÀI TOÁN

Input → Process → Output

(Dữ liệu vào → Xử lý → Kết quả ra)

Việc xác định bài toán tức là phải xác định xem ta phải giải quyết vấn đề gì?, với giả thiết nào đã cho và lời giải cần phải đạt những yêu cầu gì. Khác với bài toán thuần tuý toán học chỉ cần xác định rõ giả thiết và kết luận chứ không cần xác định yêu cầu về lời giải, đôi khi những bài toán tin học ứng dụng trong thực tế chỉ cần tìm lời giải tốt tới mức nào đó, thậm chí là tối ở mức chấp nhận được. Bởi lời giải tốt nhất đòi hỏi quá nhiều thời gian và chi phí.

Ví dụ:

Khi cài đặt các hàm số phức tạp trên máy tính. Nếu tính bằng cách khai triển chuỗi vô hạn thì độ chính xác cao hơn nhưng thời gian chậm hơn hàng tí lần so với phương pháp xấp xỉ. Trên thực tế việc tính toán luôn luôn cho phép chấp nhận một sai số nào đó nên các hàm số trong máy tính đều được tính bằng phương pháp xấp xỉ của giải tích số

Xác định đúng yêu cầu bài toán là rất quan trọng bởi nó ảnh hưởng tới cách thức giải quyết và chất lượng của lời giải. Một bài toán thực tế thường cho bởi những thông tin khá mơ hồ và晦暗, ta phải phát biểu lại một cách chính xác và chặt chẽ để hiểu đúng bài toán.

Ví dụ:

Bài toán: Một dự án có n người tham gia thảo luận, họ muốn chia thành các nhóm và mỗi nhóm thảo luận riêng về một phần của dự án. Nhóm có bao nhiêu người thì được trình lên bấy nhiêu ý kiến. Nếu lấy ở mỗi nhóm một ý kiến đem ghép lại thì được một bộ ý kiến triển khai dự án. Hãy tìm cách chia để số bộ ý kiến cuối cùng thu được là lớn nhất.

Phát biểu lại: Cho một số nguyên dương n, tìm các phân tích n thành tổng các số nguyên dương sao cho tích của các số đó là lớn nhất.

Trên thực tế, ta nên xét một vài trường hợp cụ thể để thông qua đó hiểu được bài toán rõ hơn và thấy được các thao tác cần phải tiến hành. Đối với những bài toán đơn giản, đôi khi chỉ cần qua ví dụ là ta đã có thể đưa về một bài toán quen thuộc để giải.

1.2. TÌM CẤU TRÚC DỮ LIỆU BIỂU DIỄN BÀI TOÁN

Khi giải một bài toán, ta cần phải định nghĩa tập hợp dữ liệu để biểu diễn tình trạng cụ thể. Việc lựa chọn này tùy thuộc vào vấn đề cần giải quyết và những thao tác sẽ tiến hành trên dữ liệu vào. Có những thuật toán chỉ thích ứng với một cách tổ chức dữ liệu nhất định, đôi với những cách tổ chức dữ liệu khác thì sẽ kém hiệu quả hoặc không thể thực hiện được. Chính vì vậy nên bước xây dựng cấu trúc dữ liệu không thể tách rời bước tìm kiếm thuật toán giải quyết vấn đề.

Các tiêu chuẩn khi lựa chọn cấu trúc dữ liệu

- ❖ Cấu trúc dữ liệu trước hết phải biểu diễn được đầy đủ các thông tin nhập và xuất của bài toán
 - ❖ Cấu trúc dữ liệu phải phù hợp với các thao tác của thuật toán mà ta lựa chọn để giải quyết bài toán.
 - ❖ Cấu trúc dữ liệu phải cài đặt được trên máy tính với ngôn ngữ lập trình đang sử dụng
- Đối với một số bài toán, trước khi tổ chức dữ liệu ta phải viết một đoạn chương trình nhỏ để **khảo sát** xem dữ liệu cần lưu trữ lớn tới mức độ nào.

1.3. TÌM THUẬT TOÁN

Thuật toán là một hệ thống chặt chẽ và rõ ràng các quy tắc nhằm xác định một dãy thao tác trên cấu trúc dữ liệu sao cho: Với một bộ dữ liệu vào, sau một số hữu hạn bước thực hiện các thao tác đã chỉ ra, ta đạt được mục tiêu đã định.

Các đặc trưng của thuật toán

1.3.1. Tính đơn nghĩa

Ở mỗi bước của thuật toán, các thao tác phải hết sức rõ ràng, không gây nên sự nhầm lẫn, lộn xộn, tuỳ tiện, đa nghĩa.

Không nên lẫn lộn tính đơn nghĩa và tính đơn định: Người ta phân loại thuật toán ra làm hai loại: Đơn định (Deterministic) và Ngẫu nhiên (Randomized). Với hai bộ dữ liệu giống nhau cho trước làm input, thuật toán đơn định sẽ thi hành các mã lệnh giống nhau và cho kết quả giống nhau, còn thuật toán ngẫu nhiên có thể thực hiện theo những mã lệnh khác nhau và cho kết quả khác nhau. Ví dụ như yêu cầu chọn một số tự nhiên x : $a \leq x \leq b$, nếu ta viết $x := a$ hay $x := b$ hay $x := (a + b) \text{ div } 2$, thuật toán sẽ luôn cho một giá trị duy nhất với dữ liệu vào là hai số tự nhiên a và b . Nhưng nếu ta viết $x := a + \text{Random}(b - a + 1)$ thì sẽ có thể thu được các kết quả khác nhau trong mỗi lần thực hiện với input là a và b tùy theo máy tính và bộ tạo số ngẫu nhiên.

1.3.2. Tính dừng

Thuật toán không được rơi vào quá trình vô hạn, phải dừng lại và cho kết quả sau một số hữu hạn bước.

1.3.3. Tính đúng

Sau khi thực hiện tất cả các bước của thuật toán theo đúng quá trình đã định, ta phải được kết quả mong muốn với mọi bộ dữ liệu đầu vào. Kết quả đó được kiểm chứng bằng yêu cầu bài toán.

1.3.4. Tính phổ dụng

Thuật toán phải dễ sửa đổi để thích ứng được với bất kỳ bài toán nào trong một lớp các bài toán và có thể làm việc trên các dữ liệu khác nhau.

1.3.5. Tính khả thi

- ❖ Kích thước phải đủ nhỏ: Ví dụ: Một thuật toán sẽ có tính hiệu quả bằng 0 nếu lượng bộ nhớ mà nó yêu cầu vượt quá khả năng lưu trữ của hệ thống máy tính.
- ❖ Thuật toán phải chuyển được thành chương trình: Ví dụ một thuật toán yêu cầu phải biểu diễn được số vô tỉ với độ chính xác tuyệt đối là không hiện thực với các hệ thống máy tính hiện nay
- ❖ Thuật toán phải được máy tính thực hiện trong thời gian cho phép, điều này khác với lời giải toán (Chỉ cần chứng minh là kết thúc sau hữu hạn bước). Ví dụ như xếp thời khoá biểu cho một học kỳ thì không thể cho máy tính chạy tới học kỳ sau mới ra được.

Ví dụ:

Input: 2 số nguyên tự nhiên a và b không đồng thời bằng 0

Output: Ước số chung lớn nhất của a và b


Thuật toán sẽ tiến hành được mô tả như sau: (Thuật toán Euclide)

Bước 1 (Input): Nhập a và b: Số tự nhiên

Bước 2: Nếu $b \neq 0$ thì chuyển sang bước 3, nếu không thì bỏ qua bước 3, đi làm bước 4

Bước 3: Đặt $r := a \bmod b$; Đặt $a := b$; Đặt $b := r$; Quay trở lại bước 2.

Bước 4 (Output): Kết luận ước số chung lớn nhất phải tìm là giá trị của a. Kết thúc thuật toán.



Hình 4: Lưu đồ thuật giải (Flowchart)

Khi mô tả thuật toán bằng ngôn ngữ tự nhiên, ta không cần phải quá chi tiết các bước và trình thực hiện mà chỉ cần mô tả một cách hình thức đủ để chuyển thành ngôn ngữ lập trình. Viết sơ đồ các thuật toán để quy là một ví dụ.

Đối với những thuật toán phức tạp và nặng về tính toán, các bước và các công thức nên mô tả một cách tường minh và chú thích rõ ràng để khi lập trình ta có thể nhanh chóng tra cứu.

Đối với những thuật toán kinh điển thì phải thuộc. Khi giải một bài toán lớn trong một thời gian giới hạn, ta chỉ phải thiết kế tổng thể còn những chỗ đã thuộc thì cứ việc lấp ráp vào.

Tính đúng đắn của những mô-đun đã thuộc ta không cần phải quan tâm nữa mà tập trung giải quyết các phần khác.

1.4. LẬP TRÌNH

Sau khi đã có thuật toán, ta phải tiến hành lập trình thể hiện thuật toán đó. Muốn lập trình đạt hiệu quả cao, cần phải có kỹ thuật lập trình tốt. Kỹ thuật lập trình tốt thể hiện ở kỹ năng viết chương trình, khả năng gõ rối và thao tác nhanh. Lập trình tốt không phải chỉ cần nắm vững ngôn ngữ lập trình là đủ, phải biết cách viết chương trình uyển chuyển, khôn khéo và phát triển dần dần để chuyển các ý tưởng ra thành chương trình hoàn chỉnh. Kinh nghiệm cho thấy một thuật toán hay nhưng do cài đặt vụng về nên khi chạy lại cho kết quả sai hoặc tốc độ chậm.

Thông thường, ta không nên cụ thể hóa ngay toàn bộ chương trình mà nên tiến hành theo phương pháp tinh chế từng bước (Stepwise refinement):

- ❖ Ban đầu, chương trình được thể hiện bằng ngôn ngữ tự nhiên, thể hiện thuật toán với các bước tổng thể, mỗi bước nêu lên một công việc phải thực hiện.
- ❖ Một công việc đơn giản hoặc là một đoạn chương trình đã được học thuộc thì ta tiến hành viết mã lệnh ngay bằng ngôn ngữ lập trình.
- ❖ Một công việc phức tạp thì ta lại chia ra thành những công việc nhỏ hơn để lại tiếp tục với những công việc nhỏ hơn đó.

Trong quá trình tinh chế từng bước, ta phải đưa ra những biểu diễn dữ liệu. Như vậy cùng với sự tinh chế các công việc, dữ liệu cũng được tinh chế dần, có cấu trúc hơn, thể hiện rõ hơn mối liên hệ giữa các dữ liệu.

Phương pháp tinh chế từng bước là một thể hiện của tư duy giải quyết vấn đề từ trên xuống, giúp cho người lập trình có được một định hướng thể hiện trong phong cách viết chương trình. Tránh việc mò mẫm, xoá đi viết lại nhiều lần, biến chương trình thành tờ giấy nháp.

1.5. KIỂM THỬ

1.5.1. Chạy thử và tìm lỗi

Chương trình là do con người viết ra, mà đã là con người thì ai cũng có thể nhầm lẫn. Một chương trình viết xong chưa chắc đã chạy được ngay trên máy tính để cho ra kết quả mong muốn. Kỹ năng tìm lỗi, sửa lỗi, điều chỉnh lại chương trình cũng là một kỹ năng quan trọng của người lập trình. Kỹ năng này chỉ có được bằng kinh nghiệm tìm và sửa chữa lỗi của chính mình.

Có ba loại lỗi:

- ❖ **Lỗi cú pháp:** Lỗi này hay gặp nhất nhưng lại dễ sửa nhất, chỉ cần nắm vững ngôn ngữ lập trình là đủ. Một người được coi là không biết lập trình nếu không biết sửa lỗi cú pháp.
- ❖ **Lỗi cài đặt:** Việc cài đặt thể hiện không đúng thuật toán đã định, đối với lỗi này thì phải xem lại tổng thể chương trình, kết hợp với các chức năng gõ rối để sửa lại cho đúng.

❖ Lỗi thuật toán: Lỗi này ít gặp nhất nhưng nguy hiểm nhất, nếu nhẹ thì phải điều chỉnh lại thuật toán, nếu nặng thì có khi phải loại bỏ hoàn toàn thuật toán sai và làm lại từ đầu.

1.5.2. Xây dựng các bộ test

Có nhiều chương trình rất khó kiểm tra tính đúng đắn. Nhất là khi ta không biết kết quả đúng là thế nào?. Vì vậy nếu như chương trình vẫn chạy ra kết quả (không biết đúng sai thế nào) thì việc tìm lỗi rất khó khăn. Khi đó ta nên làm các bộ test để thử chương trình của mình.

Các bộ test nên đặt trong các file văn bản, bởi việc tạo một file văn bản rất nhanh và mỗi lần chạy thử chỉ cần thay tên file dữ liệu vào là xong, không cần gõ lại bộ test từ bàn phím. Kinh nghiệm làm các bộ test là:

Bắt đầu với một bộ test nhỏ, đơn giản, làm bằng tay cũng có được đáp số để so sánh với kết quả chương trình chạy ra.

Tiếp theo vẫn là các bộ test nhỏ, nhưng chứa các giá trị đặc biệt hoặc tầm thường. Kinh nghiệm cho thấy đây là những test dễ sai nhất.

Các bộ test phải đa dạng, tránh sự lặp đi lặp lại các bộ test tương tự.

Có một vài test lớn chỉ để kiểm tra tính chịu đựng của chương trình mà thôi. Kết quả có đúng hay không thì trong đa số trường hợp, ta không thể kiểm chứng được với test này.

Lưu ý rằng chương trình chạy qua được hết các test không có nghĩa là chương trình đó đã đúng. Bởi có thể ta chưa xây dựng được bộ test làm cho chương trình chạy sai. Vì vậy nếu có thể, ta nên tìm cách chứng minh tính đúng đắn của thuật toán và chương trình, điều này thường rất khó.

1.6. TỐI ƯU CHƯƠNG TRÌNH

Một chương trình đã chạy đúng không có nghĩa là việc lập trình đã xong, ta phải sửa đổi lại một vài chi tiết để chương trình có thể chạy nhanh hơn, hiệu quả hơn. Thông thường, trước khi kiểm thử thì ta nên đặt mục tiêu viết chương trình sao cho đơn giản, **miễn sao chạy ra kết quả đúng** là được, sau đó khi tối ưu chương trình, ta xem lại những chỗ nào viết chưa tốt thì tối ưu lại mã lệnh để chương trình ngắn hơn, chạy nhanh hơn. Không nên viết tới đâu tối ưu mã đến đó, bởi chương trình có mã lệnh tối ưu thường phức tạp và khó kiểm soát.

Việc tối ưu chương trình nên dựa trên các tiêu chuẩn sau:

1.6.1. Tính tin cậy

Chương trình phải chạy đúng như dự định, mô tả đúng một giải thuật đúng. Thông thường khi viết chương trình, ta luôn có thói quen kiểm tra tính đúng đắn của các bước mỗi khi có thể.

1.6.2. Tính uyển chuyển

Chương trình phải dễ sửa đổi. Bởi ít có chương trình nào viết ra đã hoàn hảo ngay được mà vẫn cần phải sửa đổi lại. Chương trình viết dễ sửa đổi sẽ làm giảm bớt công sức của lập trình viên khi phát triển chương trình.

1.6.3. Tính trong sáng

Chương trình viết ra phải dễ đọc dễ hiểu, để sau một thời gian dài, khi đọc lại còn hiểu mình làm cái gì?. Để nếu có điều kiện thì còn có thể sửa sai (nếu phát hiện lỗi mới), cải tiến hay biến đổi để được chương trình giải quyết bài toán khác. Tính trong sáng của chương trình phụ thuộc rất nhiều vào công cụ lập trình và phong cách lập trình.

1.6.4. Tính hữu hiệu

Chương trình phải chạy nhanh và ít tốn bộ nhớ, tức là tiết kiệm được cả về không gian và thời gian. Để có một chương trình hữu hiệu, cần phải có giải thuật tốt và những tiêu xảo khi lập trình. Tuy nhiên, việc áp dụng quá nhiều tiêu xảo có thể khiến chương trình trở nên rối rắm, khó hiểu khi sửa đổi. Tiêu chuẩn hữu hiệu nên dừng lại ở mức chấp nhận được, không quan trọng bằng ba tiêu chuẩn trên. Bởi phần cứng phát triển rất nhanh, yêu cầu hữu hiệu không cần phải đặt ra quá nặng.

Từ những phân tích ở trên, chúng ta nhận thấy rằng việc làm ra một chương trình đòi hỏi rất nhiều công đoạn và tiêu tốn khá nhiều công sức. Chỉ một công đoạn không hợp lý sẽ làm tăng chi phí viết chương trình. Nghĩ ra cách giải quyết vấn đề đã khó, biến ý tưởng đó thành hiện thực cũng không dễ chút nào.

Những cấu trúc dữ liệu và giải thuật đề cập tới trong chuyên đề này là những kiến thức rất phổ thông, một người học lập trình không sớm thì muộn cũng phải biết tới. Chỉ hy vọng rằng khi học xong chuyên đề này, qua những cấu trúc dữ liệu và giải thuật hết sức mẫu mực, chúng ta rút ra được bài học kinh nghiệm: **Đừng bao giờ viết chương trình khi mà chưa suy xét kỹ về giải thuật và những dữ liệu cần thao tác**, bởi như vậy ta dễ mắc phải hai sai lầm trầm trọng: hoặc là sai về giải thuật, hoặc là giải thuật không thể triển khai nổi trên một cấu trúc dữ liệu không phù hợp. Chỉ cần mắc một trong hai lỗi đó thôi thì nguy cơ sụp đổ toàn bộ chương trình là hoàn toàn có thể, càng cố chữa càng bị rối, khả năng hẫu như chắc chắn là phải làm lại từ đầu^(*).

^(*) Tất nhiên, cẩn thận đến đâu thì cũng có xác suất rủi ro nhất định, ta hiểu được mức độ tai hại của hai lỗi này để hạn chế nó càng nhiều càng tốt

§2. PHÂN TÍCH THỜI GIAN THỰC HIỆN GIẢI THUẬT

2.1. GIỚI THIỆU

Với một bài toán không chỉ có một giải thuật. Chọn một giải thuật đưa tới kết quả nhanh nhất là một đòi hỏi thực tế. Như vậy cần có một căn cứ nào đó để nói rằng giải thuật này nhanh hơn giải thuật kia ?.

Thời gian thực hiện một giải thuật bằng chương trình máy tính phụ thuộc vào rất nhiều yếu tố. Một yếu tố cần chú ý nhất đó là kích thước của dữ liệu đưa vào. Dữ liệu càng lớn thì thời gian xử lý càng chậm, chẳng hạn như thời gian sắp xếp một dãy số phải chịu ảnh hưởng của số lượng các số thuộc dãy số đó. Nếu gọi n là kích thước dữ liệu đưa vào thì thời gian thực hiện của một giải thuật có thể biểu diễn một cách tương đối như một hàm của n : $T(n)$.


Phần cứng máy tính, ngôn ngữ viết chương trình và chương trình dịch ngôn ngữ ấy đều ảnh hưởng tới thời gian thực hiện. Những yếu tố này không giống nhau trên các loại máy, vì vậy không thể dựa vào chúng khi xác định $T(n)$. Tức là $T(n)$ không thể biểu diễn bằng đơn vị thời gian giờ, phút, giây được. Tuy nhiên, không phải vì thế mà không thể so sánh được các giải thuật về mặt tốc độ. Nếu như thời gian thực hiện một giải thuật là $T_1(n) = n^2$ và thời gian thực hiện của một giải thuật khác là $T_2(n) = 100n$ thì khi n đủ lớn, thời gian thực hiện của giải thuật T_2 rõ ràng nhanh hơn giải thuật T_1 . Khi đó, nếu nói rằng thời gian thực hiện giải thuật tỉ lệ thuận với n hay tỉ lệ thuận với n^2 cũng cho ta một cách đánh giá tương đối về tốc độ thực hiện của giải thuật đó khi n khá lớn. Cách đánh giá thời gian thực hiện giải thuật độc lập với máy tính và các yếu tố liên quan tới máy tính như vậy sẽ dẫn tới khái niệm gọi là **độ phức tạp tính toán của giải thuật**.

2.2. CÁC KÝ PHÁP ĐỂ ĐÁNH GIÁ ĐỘ PHỨC TẠP TÍNH TOÁN

Cho một giải thuật thực hiện trên dữ liệu với kích thước n . Giả sử $T(n)$ là thời gian thực hiện một giải thuật đó, $g(n)$ là một hàm xác định dương với mọi n . Khi đó ta nói độ phức tạp tính toán của giải thuật là:

- ❖ $\Theta(g(n))$ nếu tồn tại các hằng số dương c_1, c_2 và n_0 sao cho $c_1.g(n) \leq f(n) \leq c_2.g(n)$ với mọi $n \geq n_0$. Ký pháp này được gọi là ký pháp Θ lớn (big-theta notation). Trong ký pháp Θ lớn, hàm $g(.)$ được gọi là giới hạn chặt (asymptotically tight bound) của hàm $T(.)$
- ❖ $O(g(n))$ nếu tồn tại các hằng số dương c và n_0 sao cho $T(n) \leq c.g(n)$ với mọi $n \geq n_0$. Ký pháp này được gọi là ký pháp chữ O lớn (big-oh notation). Trong ký pháp chữ O lớn, hàm $g(.)$ được gọi là giới hạn trên (asymptotic upper bound) của hàm $T(.)$
- ❖ $\Omega(g(n))$ nếu tồn tại các hằng số dương c và n_0 sao cho $c.g(n) \leq T(n)$ với mọi $n \geq n_0$. Ký hiệu này gọi là ký pháp Ω lớn (big-omega notation). Trong ký pháp Ω lớn, hàm $g(.)$ được gọi là giới hạn dưới (asymptotic lower bound) của hàm $T(.)$

Hình 5 là biểu diễn đồ thị của ký pháp Θ lớn, O lớn và Ω lớn. Để thấy rằng $T(n) = \Theta(g(n))$ nếu và chỉ nếu $T(n) = O(g(n))$ và $T(n) = \Omega(g(n))$.



Hình 5: Ký pháp Θ lớn, O lớn và Ω lớn

Ta nói độ phức tạp tính toán của giải thuật là

- ❖ $o(g(n))$ nếu với mọi hằng số dương c , tồn tại một hằng số dương n_0 sao cho $T(n) \leq c.g(n)$ với mọi $n \geq n_0$. Ký pháp này gọi là kí pháp chữ o nhỏ (little-oh notation).
- ❖ $\omega(g(n))$ nếu với mọi hằng số dương c , tồn tại một hằng số dương n_0 sao cho $c.g(n) \leq T(n)$ với mọi $n \geq n_0$. Ký pháp này gọi là kí pháp ω nhỏ (little-omega notation)

Ví dụ nếu $T(n) = n^2 + 1$, thì:

- ❖ $T(n) = O(n^2)$. Thật vậy, chọn $c = 2$ và $n_0 = 1$. Rõ ràng với mọi $n \geq 1$, ta có:

$$T(n) = n^2 + 1 \leq 2n^2 = 2.g(n)$$
- ❖ $T(n) \neq o(n^2)$. Thật vậy, chọn $c = 1$. Rõ ràng không tồn tại n để: $n^2 + 1 \leq n^2$, tức là không tồn tại n_0 thoả mãn định nghĩa của kí pháp chữ o nhỏ.

Lưu ý rằng không có kí pháp θ nhỏ

Một vài tính chất:

Tính bắc cầu (transitivity): Tất cả các kí pháp nêu trên đều có tính bắc cầu:

- ❖ Nếu $f(n) = \Theta(g(n))$ và $g(n) = \Theta(h(n))$ thì $f(n) = \Theta(h(n))$.
- ❖ Nếu $f(n) = O(g(n))$ và $g(n) = O(h(n))$ thì $f(n) = O(h(n))$.
- ❖ Nếu $f(n) = \Omega(g(n))$ và $g(n) = \Omega(h(n))$ thì $f(n) = \Omega(h(n))$.
- ❖ Nếu $f(n) = o(g(n))$ và $g(n) = o(h(n))$ thì $f(n) = o(h(n))$.
- ❖ Nếu $f(n) = \omega(g(n))$ và $g(n) = \omega(h(n))$ thì $f(n) = \omega(h(n))$.

Tính phản xạ (reflexivity): Chỉ có các kí pháp “lớn” mới có tính phản xạ:

- ❖ $f(n) = \Theta(f(n))$.
- ❖ $f(n) = O(f(n))$.
- ❖ $f(n) = \Omega(f(n))$.

Tính đối xứng (symmetry): Chỉ có kí pháp Θ lớn có tính đối xứng:

❖ $f(n) = \Theta(g(n))$ nếu và chỉ nếu $g(n) = \Theta(f(n))$.

Tính chuyển vị đổi xứng (transpose symmetry):

❖ $f(n) = O(g(n))$ nếu và chỉ nếu $g(n) = \Omega(f(n))$.

❖ $f(n) = o(g(n))$ nếu và chỉ nếu $g(n) = \omega(f(n))$.

Để dễ nhớ ta coi các ký pháp $O, \Omega, \Theta, o, \omega$ lần lượt tương ứng với các phép so sánh $\leq, \geq, =, <, >$. Từ đó suy ra các tính chất trên.

2.3. XÁC ĐỊNH ĐỘ PHÚC TẠP TÍNH TOÁN CỦA GIẢI THUẬT

Việc xác định độ phức tạp tính toán của một giải thuật bất kỳ có thể rất phức tạp. Tuy nhiên độ phức tạp tính toán của một số giải thuật trong thực tế có thể tính bằng một số qui tắc đơn giản.

2.3.1. Qui tắc bỏ hằng số

Nếu đoạn chương trình P có thời gian thực hiện $T(n) = O(c_1.f(n))$ với c_1 là một hằng số dương thì có thể coi đoạn chương trình đó có độ phức tạp tính toán là $O(f(n))$.

Chứng minh:

$T(n) = O(c_1.f(n))$ nên $\exists c_0 > 0$ và $\exists n_0 > 0$ để $T(n) \leq c_0.c_1.f(n)$ với $\forall n \geq n_0$. Đặt $C = c_0.c_1$ và dùng định nghĩa, ta có $T(n) = O(f(n))$.

Qui tắc này cũng đúng với các ký pháp Ω, Θ, o và ω .

2.3.2. Quy tắc lấy max

Nếu đoạn chương trình P có thời gian thực hiện $T(n) = O(f(n) + g(n))$ thì có thể coi đoạn chương trình đó có độ phức tạp tính toán $O(\max(f(n), g(n)))$.

Chứng minh

$T(n) = O(f(n) + g(n))$ nên $\exists C > 0$ và $\exists n_0 > 0$ để $T(n) \leq C.f(n) + C.g(n)$, $\forall n \geq n_0$.

Vậy $T(n) \leq C.f(n) + C.g(n) \leq 2C.\max(f(n), g(n))$ ($\forall n \geq n_0$).

Từ định nghĩa suy ra $T(n) = O(\max(f(n), g(n)))$.

Quy tắc này cũng đúng với các ký pháp Ω, Θ, o và ω .

2.3.3. Quy tắc cộng

Nếu đoạn chương trình P_1 có thời gian thực hiện $T_1(n) = O(f(n))$ và đoạn chương trình P_2 có thời gian thực hiện là $T_2(n) = O(g(n))$ thì thời gian thực hiện P_1 rồi đến P_2 tiếp theo sẽ là

$$T_1(n) + T_2(n) = O(f(n) + g(n))$$

Chứng minh:

$T_1(n) = O(f(n))$ nên $\exists n_1 > 0$ và $c_1 > 0$ để $T_1(n) \leq c_1.f(n)$ với $\forall n \geq n_1$.

$T_2(n) = O(g(n))$ nên $\exists n_2 > 0$ và $c_2 > 0$ để $T_2(n) \leq c_2.g(n)$ với $\forall n \geq n_2$.

Chọn $n_0 = \max(n_1, n_2)$ và $c = \max(c_1, c_2)$ ta có:

Với $\forall n \geq n_0$:

$T_1(n) + T_2(n) \leq c_1.f(n) + c_2.g(n) \leq c.f(n) + c.g(n) \leq c.(f(n) + g(n))$

Vậy $T_1(n) + T_2(n) = O(f(n) + g(n))$.

Quy tắc cộng cũng đúng với các ký pháp Ω , Θ , ω và ω .

2.3.4. Quy tắc nhân

Nếu đoạn chương trình P có thời gian thực hiện là $T(n) = O(f(n))$. Khi đó, nếu thực hiện k(n) lần đoạn chương trình P với $k(n) = O(g(n))$ thì độ phức tạp tính toán sẽ là $O(g(n).f(n))$

Chứng minh:

Thời gian thực hiện $k(n)$ lần đoạn chương trình P sẽ là $k(n)T(n)$. Theo định nghĩa:

$\exists c_k \geq 0$ và $n_k > 0$ để $k(n) \leq c_k(g(n))$ với $\forall n \geq n_k$

$\exists c_T \geq 0$ và $n_T > 0$ để $T(n) \leq c_T(f(n))$ với $\forall n \geq n_T$

Vậy với $\forall n \geq \max(n_T, n_k)$ ta có $k(n).T(n) \leq c_T.c_k(g(n).f(n))$

Quy tắc nhân cũng đúng với các ký pháp Ω , Θ , ω và ω .

2.3.5. Định lý Master (Master Theorem)

Cho $a \geq 1$ và $b > 1$ là hai hằng số, $f(n)$ là một hàm với đối số n , $T(n)$ là một hàm xác định trên tập các số tự nhiên được định nghĩa như sau:

$$T(n) = a.T(n/b) + f(n)$$

Ở đây n/b có thể hiểu là $\lfloor n/b \rfloor$ hay $\lceil n/b \rceil$. Khi đó:

❖ Nếu $f(n) = O(n^{\log_b a - \varepsilon})$ với hằng số $\varepsilon > 0$, thì $T(n) = \Theta(n^{\log_b a})$

❖ Nếu $f(n) = \Theta(n^{\log_b a})$ thì $T(n) = \Theta(n^{\log_b a} \lg n)$

❖ Nếu $f(n) = \Omega(n^{\log_b a + \varepsilon})$ với hằng số $\varepsilon > 0$ và $a.f(n/b) \leq c.f(n)$ với hằng số $c < 1$ và n đủ lớn thì $T(n) = \Theta(f(n))$

Định lý Master là một định lý quan trọng trong việc phân tích độ phức tạp tính toán của các giải thuật lặp hay đệ quy. Tuy nhiên việc chứng minh định lý khá dài dòng, ta có thể tham khảo trong các tài liệu khác.

2.3.6. Một số tính chất

Ta quan tâm chủ yếu tới các ký pháp “lớn”. Rõ ràng ký pháp Θ là “chặt” hơn ký pháp O và Ω theo nghĩa: Nếu độ phức tạp tính toán của giải thuật có thể viết là $\Theta(f(n))$ thì cũng có thể viết là $O(f(n))$ cũng như $\Omega(f(n))$. Dưới đây là một số cách biểu diễn độ phức tạp tính toán qua ký pháp Θ .

❖ Nếu một thuật toán có thời gian thực hiện là $P(n)$, trong đó $P(n)$ là một đa thức bậc k thì độ phức tạp tính toán của thuật toán đó có thể viết là $\Theta(n^k)$.

- ❖ Nếu một thuật toán có thời gian thực hiện là $\log_a f(n)$. Với b là một số dương, ta nhận thấy $\log_a f(n) = \log_b a \cdot \log_b f(n)$. Tức là: $\Theta(\log_a f(n)) = \Theta(\log_b f(n))$. Vậy ta có thể nói rằng độ phức tạp tính toán của thuật toán đó là $\Theta(\log f(n))$ mà không cần ghi cơ số của logarit.
- ❖ Nếu một thuật toán có độ phức tạp là hằng số, tức là thời gian thực hiện không phụ thuộc vào kích thước dữ liệu vào thì ta ký hiệu độ phức tạp tính toán của thuật toán đó là $\Theta(1)$.

Dưới đây là một số hàm số hay dùng để ký hiệu độ phức tạp tính toán và bảng giá trị của chúng để tiện theo dõi sự tăng của hàm theo đối số n.

lgn	n	nlgn	n^2	n^3	2^n
0	1	0	1	1	2
1	2	2	4	8	4
2	4	8	16	64	16
3	8	24	64	512	256
4	16	64	256	4096	65536
5	32	160	1024	32768	2147483648

Ví dụ:

Thuật toán tính tổng các số từ 1 tới n:

Nếu viết theo sơ đồ như sau:

```
Input n;
S := 0;
for i := 1 to n do S := S + i;
Output S;
```

Các đoạn chương trình ở các dòng 1, 2 và 4 có độ phức tạp tính toán là $\Theta(1)$. Vòng lặp ở dòng 3 lặp n lần phép gán $S := S + i$, nên thời gian tính toán tỉ lệ thuận với n. Tức là độ phức tạp tính toán là $\Theta(n)$. Dùng quy tắc cộng và quy tắc lấy max, ta suy ra độ phức tạp tính toán của giải thuật trên là $\Theta(n)$.

Còn nếu viết theo sơ đồ như sau:

```
Input n;
S := n * (n - 1) div 2;
Output S;
```

Thì độ phức tạp tính toán của thuật toán trên là $\Theta(1)$, thời gian tính toán không phụ thuộc vào n.

2.3.7. Phép toán tích cực

Dựa vào những nhận xét đã nêu ở trên về các quy tắc khi đánh giá thời gian thực hiện giải thuật, ta chú ý đặc biệt đến một phép toán mà ta gọi là phép toán tích cực trong một đoạn chương trình. Đó là **một phép toán trong một đoạn chương trình mà số lần thực hiện không ít hơn các phép toán khác**.

Xét hai đoạn chương trình tính e^x bằng công thức gần đúng:

$$e^x \approx 1 + \frac{x}{1!} + \frac{x^2}{2!} + \dots + \frac{x^n}{n!} = \sum_{i=0}^n \frac{x^i}{i!} \text{ với } x \text{ và } n \text{ cho trước.}$$

{Chuong trình 1: Tính riêng từng hạng tử rồi cộng lại}

```
program Exp1;
var
  i, j, n: Integer;
  x, p, S: Real;
begin
  Write('x, n = '); ReadLn(x, n);
  S := 0;
  for i := 0 to n do
    begin
      p := 1;
      for j := 1 to i do p := p * x / j;
      S := S + p;
    end;
  WriteLn('exp(', x:1:4, ') = ', S:1:4);
end.
```

Ta có thể coi phép toán tích cực ở đây là:

$$p := p * x / j;$$

Số lần thực hiện phép toán này là:

$$0 + 1 + 2 + \dots + n = n(n - 1)/2 \text{ lần.}$$

Vậy độ phức tạp tính toán của thuật toán là $\Theta(n^2)$

{Tính hạng tử sau qua hạng tử trước}

```
program Exp2;
var
  i, n: Integer;
  x, p, S: Real;
begin
  Write('x, n = '); ReadLn(x, n);
  S := 1; p := 1;
  for i := 1 to n do
    begin
      p := p * x / i;
      S := S + p;
    end;
  WriteLn('exp(', x:1:4, ') = ', S:1:4);
end.
```

Ta có thể coi phép toán tích cực ở đây là:

$$p := p * x / i;$$

Số lần thực hiện phép toán này là n .

Vậy độ phức tạp tính toán của thuật toán là $\Theta(n)$.

2.4. ĐỘ PHỨC TẠP TÍNH TOÁN VỚI TÌNH TRẠNG DỮ LIỆU VÀO

Có nhiều trường hợp, thời gian thực hiện giải thuật không phải chỉ phụ thuộc vào kích thước dữ liệu mà còn phụ thuộc vào tình trạng của dữ liệu đó nữa. Chẳng hạn thời gian sắp xếp một dãy số theo thứ tự tăng dần mà dãy đưa vào chưa có thứ tự sẽ khác với thời gian sắp xếp một dãy số đã sắp xếp rồi hoặc đã sắp xếp theo thứ tự ngược lại. Lúc này, khi phân tích thời gian thực hiện giải thuật ta sẽ phải xét tới trường hợp tốt nhất, trường hợp trung bình và trường hợp xấu nhất.

- ❖ Phân tích thời gian thực hiện giải thuật trong trường hợp xấu nhất (worst-case analysis): Với một kích thước dữ liệu n , tìm $T(n)$ là thời gian lớn nhất khi thực hiện giải thuật trên mọi bộ dữ liệu kích thước n và phân tích thời gian thực hiện giải thuật dựa trên hàm $T(n)$.
- ❖ Phân tích thời gian thực hiện giải thuật trong trường hợp tốt nhất (best-case analysis): Với một kích thước dữ liệu n , tìm $T(n)$ là thời gian ít nhất khi thực hiện giải thuật trên mọi bộ dữ liệu kích thước n và phân tích thời gian thực hiện giải thuật dựa trên hàm $T(n)$.
- ❖ Phân tích thời gian trung bình thực hiện giải thuật (average-case analysis): Giả sử rằng dữ liệu vào tuân theo một phân phối xác suất nào đó (chẳng hạn phân bố đều nghĩa là khả năng chọn mỗi bộ dữ liệu vào là như nhau) và tính toán giá trị kỳ vọng (trung bình) của thời gian chạy cho mỗi kích thước dữ liệu n ($T(n)$), sau đó phân tích thời gian thực hiện giải thuật dựa trên hàm $T(n)$.

Khi khó khăn trong việc xác định độ phức tạp tính toán trung bình (bởi việc xác định $T(n)$ trung bình thường phải dùng tới những công cụ toán phức tạp), người ta thường chỉ đánh giá độ phức tạp tính toán trong trường hợp xấu nhất.

Không nên lẫn lộn các cách phân tích trong trường hợp xấu nhất, trung bình, và tốt nhất với các ký pháp biểu diễn độ phức tạp tính toán, đây là hai khái niệm hoàn toàn phân biệt.

Trên phương diện lý thuyết, đánh giá bằng ký pháp $\Theta(\cdot)$ là tốt nhất, tuy vậy việc đánh giá bằng ký pháp $\Theta(\cdot)$ đòi hỏi phải đánh giá bằng cả ký pháp $O(\cdot)$ lẫn $\Omega(\cdot)$. Dẫn tới việc phân tích khá phức tạp, gần như phải biểu diễn chính xác thời gian thực hiện giải thuật qua các hàm giải tích. Vì vậy trong những thuật toán về sau, phần lớn tôi sẽ dùng ký pháp $T(n) = O(f(n))$ với $f(n)$ là hàm tăng chậm nhất có thể (nằm trong tầm hiểu biết của mình).

2.5. CHI PHÍ THỰC HIỆN THUẬT TOÁN

Khái niệm độ phức tạp tính toán đặt ra không chỉ dùng để đánh giá chi phí thực hiện một giải thuật về mặt thời gian mà là để đánh giá chi phí thực hiện giải thuật nói chung, bao gồm cả chi phí về không gian (lượng bộ nhớ cần sử dụng). Tuy nhiên ở trên ta chỉ đưa định nghĩa về độ phức tạp tính toán dựa trên chi phí về thời gian cho dễ trình bày. Việc đánh giá độ phức tạp tính toán theo các tiêu chí khác cũng tương tự nếu ta biểu diễn được mức chi phí theo một hàm $T(\cdot)$ của kích thước dữ liệu vào. Nếu phát biểu rằng độ phức tạp tính toán của một giải thuật là $\Theta(n^2)$ về thời gian và $\Theta(n)$ về bộ nhớ cũng không có gì sai về mặt ngữ nghĩa cả.

Thông thường,

- ❖ Nếu ta đánh giá được độ phức tạp tính toán của một giải thuật qua ký pháp Θ , có thể coi phép đánh giá này là đủ chặt và không cần đánh giá qua những ký pháp khác nữa.
- ❖ Nếu không:

Để nhấn mạnh đến tính “tốt” của một giải thuật, các ký pháp O , o thường được sử dụng. Nếu đánh giá được qua O thì không cần đánh giá qua o . Ý nói: Chi phí thực hiện thuật toán tối đa là..., ít hơn...

Để đề cập đến tính “tồi” của một giải thuật, các ký pháp Ω , ω thường được sử dụng. Nếu đánh giá được qua Ω thì không cần đánh giá qua ω . Ý nói: Chi phí thực hiện thuật toán tối thiểu là..., cao hơn ...

Bài tập

Bài 1

Có 16 giải thuật với chi phí lần lượt là $g_1(n), g_2(n), \dots, g_{16}(n)$ được liệt kê dưới đây

$$2^{100^{100}}, (\sqrt{2})^{\lg n}, n^2, n!, 3^n, \sum_{k=1}^n k, \lg^* n, \lg(n!), 1, \lg^*(\lg n), \ln n, n^{\lg(\lg n)}, (\lg n)^{\lg n}, 2^n,$$

$$n \lg n, \sum_{k=1}^n \frac{1}{k}$$

Ở đây n là kích thước dữ liệu vào.

(Ta dùng ký hiệu $\lg x$ chỉ logarithm nhị phân (cơ số 2) của x chứ không phải là logarithm thập phân theo hệ thống ký hiệu của Nga. Hàm $\lg^* x$ cho giá trị bằng số lần lấy logarithm nhị phân để thu được giá trị ≤ 1 từ số x . Ví dụ: $\lg^* 2 = 1, \lg^* 4 = 2, \lg^* 16 = 3, \lg^* 65536 = 4\dots$)

Hãy xếp lại các giải thuật theo chiều tăng của độ phức tạp tính toán. Có nghĩa là tìm một thứ tự $g_{i[1]}, g_{i[2]}, \dots, g_{i[16]}$ sao cho $g_{i[1]} = O(g_{i[2]}), g_{i[2]} = O(g_{i[3]}), \dots, g_{i[15]} = O(g_{i[16]})$. Chỉ rõ các giải thuật nào là “tương đương” về độ phức tạp tính toán theo nghĩa $g_{i[j]} = \Theta(g_{i[j+1]})$.

Đáp án:

$$1, 2^{100^{100}}$$

$$\lg^* n, \lg^*(\lg n)$$

$$\ln n, \sum_{k=1}^n \frac{1}{k}$$

$$(\sqrt{2})^{\lg n}$$

$$n \lg n, \lg(n!)$$

$$n^2, \sum_{k=1}^n k$$

$$(\lg n)^{\lg n}, n^{\lg(\lg n)}$$

$$2^n$$

$$3^n$$

$$n!$$

Bài 2

Xác định độ phức tạp tính toán của những giải thuật sau bằng ký pháp Θ :

a) Đoạn chương trình tính tổng hai đa thức:

$$P(x) = a_m x^m + a_{m-1} x^{m-1} + \dots + a_1 x + a_0 \text{ và } Q(x) = b_n x^n + b_{n-1} x^{n-1} + \dots + b_1 x + b_0$$

Để được đa thức

$$R(x) = c_p x^p + c_{p-1} x^{p-1} + \dots + c_1 x + c_0$$

```
if m < n then p := m else p := n; {p = min(m, n)}
for i := 0 to p do c[i] := a[i] + b[i];
if p < m then
    for i := p + 1 to m do c[i] := a[i]
else
    for i := p + 1 to n do c[i] := b[i];
while (p > 0) and (c[p] = 0) do p := p - 1;
```

b) Đoạn chương trình tính tích hai đa thức:

$$P(x) = a_m x^m + a_{m-1} x^{m-1} + \dots + a_1 x + a_0 \text{ và } Q(x) = b_n x^n + b_{n-1} x^{n-1} + \dots + b_1 x + b_0$$

Để được đa thức

$$R(x) = c_p x^p + c_{p-1} x^{p-1} + \dots + c_1 x + c_0$$

```
p := m + n;
for i := 0 to p do c[i] := 0;
```

```

for i := 0 to m do
  for j := 0 to n do
    c[i + j] := c[i + j] + a[i] * b[j];
  
```

Đáp án

a) $\Theta(\max(m, n))$; b) $\Theta(m \cdot n)$

Bài 3

Chỉ ra rằng cách nói “Độ phức tạp tính toán của giải thuật A tối thiểu phải là $O(n^2)$ ” là không thực sự chính xác.

(ký pháp O không liên quan gì đến chuyên đánh giá “tối thiểu” cả).

Bài 4

Giải thích tại sao không có kíp pháp $\theta(f(n))$ để chỉ những hàm vừa là $\theta(f(n))$ vừa là $\omega(f(n))$.

(Vì không có hàm nào vừa là $\theta(f(n))$ vừa là $\omega(f(n))$)

Bài 5

Chứng minh rằng

$$n! = \theta(n^n)$$

$$n! = \omega(2^n)$$

$$\lg(n!) = \Theta(n \lg n)$$

Hướng dẫn: Dùng công thức xấp xỉ của Stirling: $n! = \sqrt{2\pi n} \left(\frac{n}{e}\right)^n \left(1 + \Theta\left(\frac{1}{n}\right)\right)$

Bài 5

Chỉ ra chỗ sai trong chứng minh sau

Giả sử một giải thuật có thời gian thực hiện $T(n)$ cho bởi

$$T(n) = \begin{cases} 1, & \text{if } n \leq 1 \\ 2T(\lceil n/2 \rceil) + n, & \text{if } n > 1 \end{cases}$$

Khi đó $T(n)$ là $\Omega(n \lg n)$ và $T(n)$ cũng là $O(n)!!!$

Chứng minh:

a) $T(n) = \Omega(n \lg n)$

Thật vậy, hoàn toàn có thể chọn một số dương c nhỏ hơn 1 và đủ nhỏ để $T(n) \geq c(n \lg n)$ với $\forall n < 3$ (chẳng hạn chọn $c = 0.1$). Ta sẽ chứng minh $T(n) \geq c(n \lg n)$ cũng đúng với $\forall n \geq 3$ bằng phương pháp quy nạp: nếu $T(\lceil n/2 \rceil) \geq c \lceil n/2 \rceil \lg \lceil n/2 \rceil$ thì $T(n) \geq c(n \lg n)$.

$$\begin{aligned} T(n) &= 2T(\lceil n/2 \rceil) + n \\ &\geq 2c \lceil n/2 \rceil \lg (\lceil n/2 \rceil) + n \quad (\text{Giả thiết quy nạp}) \end{aligned}$$

$$\begin{aligned}
 &\geq 2c(n/2)\lg(n/2) + n \\
 &= cn\lg(n/2) + n \\
 &= cn\lg n - cn\lg 2 + n \\
 &= cn\lg n + (1-c)n \\
 &\geq cn\lg n \quad (\text{Với cách chọn hằng số } c < 1)
 \end{aligned}$$

b) $T(n) = O(n)$

Thật vậy, ta lại có thể chọn một số dương c đủ lớn để $T(n) < cn$ với $\forall n < 3$. Ta sẽ chứng minh quy nạp cho trường hợp $n \geq 3$:

$$\begin{aligned}
 T(n) &= 2T(\lceil n/2 \rceil) + n \\
 &\leq 2c\lceil n/2 \rceil + n \quad (\text{Giả thiết quy nạp}) \\
 &\leq c(n+1) + n \quad (= \text{nhi thức bậc nhất của } n) \\
 &= O(n)
 \end{aligned}$$

Ta được một kết quả “thú vị”: Từ $nlgn = O(T(n))$ và $T(n) = O(n)$, theo luật bắc cầu ta suy ra: $nlgn = O(n)$!!!wow!!!

§3. ĐỆ QUY VÀ GIẢI THUẬT ĐỆ QUY

3.1. KHÁI NIỆM VỀ ĐỆ QUY

Ta nói một đối tượng là đệ quy nếu nó được định nghĩa qua chính nó hoặc một đối tượng khác cùng dạng với chính nó bằng quy nạp.

Ví dụ: Đặt hai chiếc gương cầu đối diện nhau. Trong chiếc gương thứ nhất chứa hình chiếc gương thứ hai. Chiếc gương thứ hai lại chứa hình chiếc gương thứ nhất nên tất nhiên nó chứa lại hình ảnh của chính nó trong chiếc gương thứ nhất... Ở một góc nhìn hợp lý, ta có thể thấy một dãy ảnh vô hạn của cả hai chiếc gương.

Một ví dụ khác là nếu người ta phát hình trực tiếp phát thanh viên ngoài bên máy vô tuyến truyền hình, trên màn hình của máy này lại có chính hình ảnh của phát thanh viên đó ngoài bên máy vô tuyến truyền hình và cứ như thế...

Trong toán học, ta cũng hay gặp các định nghĩa đệ quy:

- ❖ Giai thừa của n ($n!$): Nếu $n = 0$ thì $n! = 1$; nếu $n > 0$ thì $n! = n.(n-1)!$
- ❖ Ký hiệu số phần tử của một tập hợp hữu hạn S là $|S|$: Nếu $S = \emptyset$ thì $|S| = 0$; Nếu $S \neq \emptyset$ thì tất có một phần tử $x \in S$, khi đó $|S| = |S \setminus \{x\}| + 1$. Đây là phương pháp định nghĩa tập các số tự nhiên.

3.2. GIẢI THUẬT ĐỆ QUY

Nếu lời giải của một bài toán P được thực hiện bằng lời giải của bài toán P' có dạng giống như P thì đó là một lời giải đệ quy. Giải thuật tương ứng với lời giải như vậy gọi là giải thuật đệ quy. Mới nghe thì có vẻ hơi lạ nhưng điểm mấu chốt cần lưu ý là: P' tuy có dạng giống như P , nhưng theo một nghĩa nào đó, nó phải “nhỏ” hơn P , dễ giải hơn P và việc giải nó không cần dùng đến P .

Trong Pascal, ta đã thấy nhiều ví dụ của các hàm và thủ tục có chứa lời gọi đệ quy tới chính nó, bây giờ, ta tóm tắt lại các phép đệ quy trực tiếp và tương hỗ được viết như thế nào:

Định nghĩa một hàm đệ quy hay thủ tục đệ quy gồm hai phần:

- ❖ Phần neo (anchor): Phần này được thực hiện khi mà công việc quá đơn giản, có thể giải trực tiếp chứ không cần phải nhờ đến một bài toán con nào cả.
- ❖ Phần đệ quy: Trong trường hợp bài toán chưa thể giải được bằng phần neo, ta xác định những bài toán con và gọi đệ quy giải những bài toán con đó. Khi đã có lời giải (đáp số) của những bài toán con rồi thì phối hợp chúng lại để giải bài toán đang quan tâm.

Phần đệ quy thể hiện tính “quy nạp” của lời giải. Phần neo cũng rất quan trọng bởi nó quyết định tới tính hữu hạn dừng của lời giải.

3.3. VÍ DỤ VỀ GIẢI THUẬT ĐỆ QUY

3.3.1. Hàm tính giai thừa

```
function Factorial(n: Integer): Integer; {Nhận vào số tự nhiên n và trả về n!}
begin
  if n = 0 then Factorial := 1 {Phần neo}
  else Factorial := n * Factorial(n - 1); {Phần đệ quy}
end;
```

Ở đây, phần neo định nghĩa kết quả hàm tại $n = 0$, còn phần đệ quy (ứng với $n > 0$) sẽ định nghĩa kết quả hàm qua giá trị của n và giai thừa của $n - 1$.

Ví dụ: Dùng hàm này để tính $3!$, trước hết nó phải đi tính $2!$ bởi $3!$ được tính bằng tích của $3 * 2!$. Tương tự để tính $2!$, nó lại đi tính $1!$ bởi $2!$ được tính bằng $2 * 1!$. Áp dụng bước quy nạp này thêm một lần nữa, $1! = 1 * 0!$, và ta đạt tới trường hợp của phần neo, đến đây từ giá trị 1 của $0!$, nó tính được $1! = 1 * 1 = 1$; từ giá trị của $1!$ nó tính được $2!$; từ giá trị của $2!$ nó tính được $3!$; cuối cùng cho kết quả là 6:

$$\begin{aligned} 3! &= 3 * 2! \\ &\downarrow \\ 2! &= 2 * 1! \\ &\downarrow \\ 1! &= 1 * 0! \\ &\downarrow \\ 0! &= 1 \end{aligned}$$

3.3.2. Dãy số Fibonacci

Dãy số Fibonacci bắt nguồn từ bài toán cổ về việc sinh sản của các cặp thỏ. Bài toán đặt ra như sau:

- ❖ Các con thỏ không bao giờ chết
 - ❖ Hai tháng sau khi ra đời, mỗi cặp thỏ mới sẽ sinh ra một cặp thỏ con (một đực, một cái)
 - ❖ Khi đã sinh con rồi thì cứ mỗi tháng tiếp theo chúng lại sinh được một cặp con mới
- Giả sử từ đầu tháng 1 có một cặp mới ra đời thì đến giữa tháng thứ n sẽ có bao nhiêu cặp.

Ví dụ, $n = 5$, ta thấy:

- ❖ Giữa tháng thứ 1: 1 cặp (ab) (cặp ban đầu)
- ❖ Giữa tháng thứ 2: 1 cặp (ab) (cặp ban đầu vẫn chưa đẻ)
- ❖ Giữa tháng thứ 3: 2 cặp (AB)(cd) (cặp ban đầu đẻ ra thêm 1 cặp con)
- ❖ Giữa tháng thứ 4: 3 cặp (AB)(cd)(ef) (cặp ban đầu tiếp tục đẻ)
- ❖ Giữa tháng thứ 5: 5 cặp (AB)(CD)(ef)(gh)(ik) (cả cặp (AB) và (CD) cùng đẻ)

Bây giờ, ta xét tới việc tính số cặp thỏ ở tháng thứ n : $F(n)$

Nếu mỗi cặp thỏ ở tháng thứ $n - 1$ đều sinh ra một cặp thỏ con thì số cặp thỏ ở tháng thứ n sẽ là:

$$F(n) = 2 * F(n - 1)$$

Nhưng vấn đề không phải như vậy, trong các cặp thỏ ở tháng thứ $n - 1$, chỉ có những cặp thỏ đã có ở tháng thứ $n - 2$ mới sinh con ở tháng thứ n được thôi. Do đó $F(n) = F(n - 1) + F(n - 2)$ ($=$ số cũ + số sinh ra). Vậy có thể tính được $F(n)$ theo công thức sau:

$$F(n) = 1 \text{ nếu } n \leq 2$$

$$F(n) = F(n - 1) + F(n - 2) \text{ nếu } n > 2$$

```
function F(n: Integer): Integer; {Tính số cặp thỏ ở tháng thứ n}
begin
  if n ≤ 2 then F := 1 {Phản ứng}
  else F := F(n - 1) + F(n - 2); {Phản ứng quy}
end;
```

3.3.3. Giả thuyết của Collatz

Collatz đưa ra giả thuyết rằng: với một số nguyên dương X , nếu X chẵn thì ta gán $X := X \text{ div } 2$; nếu X lẻ thì ta gán $X := X * 3 + 1$. Thì sau một số hữu hạn bước, ta sẽ có $X = 1$.

Ví dụ: $X = 10$, các bước tiến hành như sau:

1. $X = 10$ (chẵn) $\Rightarrow X := 10 \text{ div } 2$; (5)
2. $X = 5$ (lẻ) $\Rightarrow X := 5 * 3 + 1$; (16)
3. $X = 16$ (chẵn) $\Rightarrow X := 16 \text{ div } 2$; (8)
4. $X = 8$ (chẵn) $\Rightarrow X := 8 \text{ div } 2$ (4)
5. $X = 4$ (chẵn) $\Rightarrow X := 4 \text{ div } 2$ (2)
6. $X = 2$ (chẵn) $\Rightarrow X := 2 \text{ div } 2$ (1)

Cứ cho giả thuyết Collatz là đúng đắn, vấn đề đặt ra là: Cho trước số 1 cùng với hai phép toán $* 2$ và $\text{div } 3$, hãy sử dụng một cách hợp lý hai phép toán đó để biến số 1 thành một giá trị nguyên dương X cho trước.

Ví dụ: $X = 10$ ta có $1 * 2 * 2 * 2 * 2 \text{ div } 3 * 2 = 10$.

Để thấy rằng lời giải của bài toán gần như thứ tự ngược của phép biến đổi Collatz: Để biểu diễn số $X > 1$ bằng một biểu thức bắt đầu bằng số 1 và hai phép toán “ $* 2$ ”, “ $\text{div } 3$ ”. Ta chia hai trường hợp:

Nếu X chẵn, thì ta tìm cách biểu diễn số $X \text{ div } 2$ và viết thêm phép toán $* 2$ vào cuối

Nếu X lẻ, thì ta tìm cách biểu diễn số $X * 3 + 1$ và viết thêm phép toán $\text{div } 3$ vào cuối

```
procedure Solve(X: Integer); {In ra cách biểu diễn số X}
begin
  if X = 1 then Write(X) {Phản ứng}
  else {Phản ứng quy}
    if X mod 2 = 0 then {X chẵn}
      begin
        Solve(X div 2); {Tìm cách biểu diễn số X div 2}
        Write('* 2'); {Sau đó viết thêm phép toán * 2}
      end
    else {X lẻ}
      begin
        Solve(X * 3 + 1); {Tìm cách biểu diễn số X * 3 + 1}
        Write(' div 3'); {Sau đó viết thêm phép toán div 3}
      end;
end;
```

```
end;
```

Trên đây là cách viết đệ quy trực tiếp, còn có một cách viết đệ quy tương hỗ như sau:

```
procedure Solve(X: Integer); forward; {Thủ tục tìm cách biểu diễn số X: Khai báo trước, đặc tả sau}
```

```
procedure SolveOdd(X: Integer); {Thủ tục tìm cách biểu diễn số X > 1 trong trường hợp X lẻ}
```

```
begin
```

```
    Solve(X * 3 + 1);
```

```
    Write(' div 3');
```

```
end;
```

```
procedure SolveEven(X: Integer); {Thủ tục tìm cách biểu diễn số X trong trường hợp X chẵn}
```

```
begin
```

```
    Solve(X div 2);
```

```
    Write(' * 2');
```

```
end;
```

```
procedure Solve(X: Integer); {Phần đặc tả của thủ tục Solve đã khai báo trước ở trên}
```

```
begin
```

```
    if X = 1 then Write(X)
```

```
    else
```

```
        if X mod 2 = 1 then SolveOdd(X)
```

```
        else SolveEven(X);
```

```
end;
```

Trong cả hai cách viết, để tìm biểu diễn số X theo yêu cầu chỉ cần gọi Solve(X) là xong. Tuy nhiên trong cách viết đệ quy trực tiếp, thủ tục Solve có lời gọi tới chính nó, còn trong cách viết đệ quy tương hỗ, thủ tục Solve chưa lời gọi tới thủ tục SolveOdd và SolveEven, hai thủ tục này lại chứa trong nó lời gọi ngược về thủ tục Solve.


Đối với những bài toán nêu trên, việc thiết kế các giải thuật đệ quy tương ứng khá thuận lợi vì cả hai đều thuộc dạng tính giá trị hàm mà định nghĩa quy nạp của hàm đó được xác định dễ dàng.

Nhưng không phải lúc nào phép giải đệ quy cũng có thể nhìn nhận và thiết kế dễ dàng như vậy. Thế thì vấn đề gì cần lưu tâm trong phép giải đệ quy?. Có thể tìm thấy câu trả lời qua việc giải đáp các câu hỏi sau:

- ❖ Có thể định nghĩa được bài toán dưới dạng phối hợp của những bài toán cùng loại nhưng nhỏ hơn hay không ? Khái niệm “nhỏ hơn” là thế nào ?
- ❖ Trường hợp đặc biệt nào của bài toán sẽ được coi là trường hợp tầm thường và có thể giải ngay được để đưa vào phần neo của phép giải đệ quy

3.3.4. Bài toán Tháp Hà Nội

Đây là một bài toán mang tính chất một trò chơi, tương truyền rằng tại ngôi đền Benares có ba cái cọc kim cương. Khi khai sinh ra thế giới, thượng đế đặt n cái đĩa bằng vàng chồng lên nhau theo thứ tự giảm dần của đường kính từ dưới lên, đĩa to nhất được đặt trên một chiếc cọc.

**Hình 6: Tháp Hà Nội**

Các nhà sư lần lượt chuyển các đĩa sang cọc khác theo luật:

- ❖ Khi di chuyển một đĩa, phải đặt nó vào một trong ba cọc đã cho
- ❖ Mỗi lần chỉ có thể chuyển một đĩa và phải là đĩa ở trên cùng
- ❖ Tại một vị trí, đĩa nào mới chuyển đến sẽ phải đặt lên trên cùng
- ❖ Đĩa lớn hơn không bao giờ được phép đặt lên trên đĩa nhỏ hơn (hay nói cách khác: một đĩa chỉ được đặt trên cọc hoặc đặt trên một đĩa lớn hơn)

Ngày tận thế sẽ đến khi toàn bộ chồng đĩa được chuyển sang một cọc khác.

Trong trường hợp có 2 đĩa, cách làm có thể mô tả như sau:

Chuyển đĩa nhỏ sang cọc 3, đĩa lớn sang cọc 2 rồi chuyển đĩa nhỏ từ cọc 3 sang cọc 2.

Những người mới bắt đầu có thể giải quyết bài toán một cách dễ dàng khi số đĩa là ít, nhưng họ sẽ gặp rất nhiều khó khăn khi số các đĩa nhiều hơn. Tuy nhiên, với tư duy quy nạp toán học và một máy tính thì công việc trở nên khá dễ dàng:

Có n đĩa.

- ❖ Nếu $n = 1$ thì ta chuyển đĩa duy nhất đó từ cọc 1 sang cọc 2 là xong.
- ❖ Giả sử rằng ta có phương pháp chuyển được $n - 1$ đĩa từ cọc 1 sang cọc 2, thì cách chuyển $n - 1$ đĩa từ cọc x sang cọc y ($1 \leq x, y \leq 3$) cũng tương tự.
- ❖ Giả sử rằng ta có phương pháp chuyển được $n - 1$ đĩa giữa hai cọc bất kỳ. Để chuyển n đĩa từ cọc x sang cọc y, ta gọi cọc còn lại là z ($=6 - x - y$). Coi đĩa to nhất là ... cọc, chuyển $n - 1$ đĩa còn lại từ cọc x sang cọc z, sau đó chuyển đĩa to nhất đó sang cọc y và cuối cùng lại coi đĩa to nhất đó là cọc, chuyển $n - 1$ đĩa còn lại đang ở cọc z sang cọc y chồng lên đĩa to nhất.

Cách làm đó được thể hiện trong thủ tục đệ quy dưới đây:

```
procedure Move(n, x, y: Integer); {Thủ tục chuyển n đĩa từ cọc x sang cọc y}
begin
  if n = 1 then Writeln('Chuyển 1 đĩa từ ', x, ' sang ', y)
  else {Để chuyển n > 1 đĩa từ cọc x sang cọc y, ta chia làm 3 công đoạn}
    begin
      Move(n - 1, x, 6 - x - y); {Chuyển n - 1 đĩa từ cọc x sang cọc trung gian}
      Move(1, x, y); {Chuyển đĩa to nhất từ x sang y}
      Move(n - 1, 6 - x - y, y); {Chuyển n - 1 đĩa từ cọc trung gian sang cọc y}
    end;
end;
```

Chương trình chính rất đơn giản, chỉ gồm có 2 việc: Nhập vào số n và gọi Move(n, 1, 2).

3.4. HIỆU LỤC CỦA ĐỆ QUY

Qua các ví dụ trên, ta có thể thấy đệ quy là một công cụ mạnh để giải các bài toán. Có những bài toán mà bên cạnh giải thuật đệ quy vẫn có những giải thuật lặp khá đơn giản và hữu hiệu. Chẳng hạn bài toán tính giai thừa hay tính số Fibonacci. Tuy vậy, đệ quy vẫn có vai trò xứng đáng của nó, có nhiều bài toán mà việc thiết kế giải thuật đệ quy đơn giản hơn nhiều so với lời giải lặp và trong một số trường hợp chương trình đệ quy hoạt động nhanh hơn chương trình viết không có đệ quy. Giải thuật cho bài Tháp Hà Nội và thuật toán sắp xếp kiểu phân đoạn (QuickSort) mà ta sẽ nói tới trong các bài sau là những ví dụ.

Có một mối quan hệ khăng khít giữa đệ quy và quy nạp toán học. Cách giải đệ quy cho một bài toán dựa trên việc định rõ lời giải cho trường hợp suy biến (neo) rồi thiết kế làm sao để lời giải của bài toán được suy ra từ lời giải của bài toán nhỏ hơn cùng loại như thế. Tương tự như vậy, quy nạp toán học chứng minh một tính chất nào đó ứng với số tự nhiên cũng bằng cách chứng minh tính chất đó đúng với một số trường hợp cơ sở (thường người ta chứng minh nó đúng với 0 hay đúng với 1) và sau đó chứng minh tính chất đó sẽ đúng với n bất kỳ nếu nó đã đúng với mọi số tự nhiên nhỏ hơn n .

Do đó ta không lấy làm ngạc nhiên khi thấy quy nạp toán học được dùng để chứng minh các tính chất có liên quan tới giải thuật đệ quy. Chẳng hạn: Chứng minh số phép chuyển đĩa để giải bài toán Tháp Hà Nội với n đĩa là $2^n - 1$:

Rõ ràng là tính chất này đúng với $n = 1$, bởi ta cần $2^1 - 1 = 1$ lần chuyển đĩa để thực hiện yêu cầu

Với $n > 1$; Giả sử rằng để chuyển $n - 1$ đĩa giữa hai cọc ta cần $2^{n-1} - 1$ phép chuyển đĩa, khi đó để chuyển n đĩa từ cọc x sang cọc y, nhìn vào giải thuật đệ quy ta có thể thấy rằng trong trường hợp này nó cần $(2^{n-1} - 1) + 1 + (2^{n-1} - 1) = 2^n - 1$ phép chuyển đĩa. Tính chất được chứng minh đúng với n

Vậy thì công thức này sẽ đúng với mọi n .

Thật đáng tiếc nếu như chúng ta phải lập trình với một công cụ không cho phép đệ quy, nhưng như vậy không có nghĩa là ta bó tay trước một bài toán mang tính đệ quy. Mọi giải thuật đệ quy đều có cách thay thế bằng một giải thuật không đệ quy (khử đệ quy), có thể nói được như vậy bởi tất cả các chương trình con đệ quy sẽ đều được trình dịch chuyển thành những mã lệnh không đệ quy trước khi giao cho máy tính thực hiện.

Việc tìm hiểu cách khử đệ quy một cách “máy móc” như các chương trình dịch thì chỉ cần hiểu rõ cơ chế xếp chồng của các thủ tục trong một dây chuyền gọi đệ quy là có thể làm được. Nhưng muốn khử đệ quy một cách tinh tế thì phải tuỳ thuộc vào từng bài toán mà khử đệ quy cho khéo. Không phải tìm đâu xa, những kỹ thuật giải công thức truy hồi bằng quy hoạch động là ví dụ cho thấy tính nghệ thuật trong những cách tiếp cận bài toán mang bản chất đệ quy để tìm ra một giải thuật không đệ quy đầy hiệu quả.

Bài tập

Bài 1

Viết một hàm đệ quy tính ước số chung lớn nhất của hai số tự nhiên a, b không đồng thời bằng 0, chỉ rõ đâu là phần neo, đâu là phần đệ quy.

Bài 2

Viết một hàm đệ quy tính $\binom{n}{k}$ theo công thức truy hồi sau:

$$\begin{cases} \binom{n}{0} = \binom{n}{n} = 1 \\ \binom{n}{k} = \binom{n-1}{k-1} + \binom{n-1}{k}; \forall k: 0 < k < n \end{cases}$$

(Ở đây tôi dùng ký hiệu $\binom{n}{k}$ thay cho C_n^k thuộc hệ thống ký hiệu của Nga)

Bài 3

Nêu rõ các bước thực hiện của giải thuật cho bài Tháp Hà Nội trong trường hợp n = 3.

Viết chương trình giải bài toán Tháp Hà Nội không đệ quy

Lời giải:

Có nhiều cách giải, ở đây tôi viết một cách “lạ” nhất với mục đích giải trí, các bạn tự tìm hiểu tại sao nó hoạt động đúng:

```
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program HanoiTower;
const
  max = 64;
var
  Stack: array[1..3, 0..max] of Integer;
  nd: array[1..3] of Integer;
  RotatedList: array[0..2, 1..2] of Integer;
  n: Integer;
  i: LongWord;

procedure Init;
var
  i: Integer;
begin
  Stack[1, 0] := n + 1; Stack[2, 0] := n + 1; Stack[3, 0] := n + 1;
  for i := 1 to n do Stack[1, i] := n + 1 - i;
  nd[1] := n; nd[2] := 0; nd[3] := 0;
  if Odd(n) then
    begin
      RotatedList[0][1] := 1; RotatedList[0][2] := 2;
      RotatedList[1][1] := 1; RotatedList[1][2] := 3;
      RotatedList[2][1] := 2; RotatedList[2][2] := 3;
    end
  else
    begin
      RotatedList[0][1] := 1; RotatedList[0][2] := 3;
      RotatedList[1][1] := 1; RotatedList[1][2] := 2;
      RotatedList[2][1] := 2; RotatedList[2][2] := 3;
    end;
end;
```

```
end;

procedure DisplayStatus;
var
  i: Integer;
begin
  for i := 1 to 3 do
    Writeln('Peg ', i, ': ', nd[i], ' disks');
end;

procedure MoveDisk(x, y: Integer);
begin
  if Stack[x][nd[x]] < Stack[y][nd[y]] then
  begin
    Writeln('Move one disk from ', x, ' to ', y);
    Stack[y][nd[y] + 1] := Stack[x][nd[x]];
    Inc(nd[y]);
    Dec(nd[x]);
  end
  else
  begin
    Writeln('Move one disk from ', y, ' to ', x);
    Stack[x][nd[x] + 1] := Stack[y][nd[y]];
    Inc(nd[x]);
    Dec(nd[y]);
  end;
end;

begin
  Write('n = '); Readln(n);
  Init;
  DisplayStatus;
  for i := 1 to LongWord(1) shl (n - 1) - 1 + LongWord(1) shl (n - 1) do
    MoveDisk(RotatedList[(i - 1) mod 3][1], RotatedList[(i - 1) mod 3][2]);
  DisplayStatus;
end.
```

§4. CẤU TRÚC DỮ LIỆU BIỂU DIỄN DANH SÁCH

4.1. KHÁI NIỆM DANH SÁCH

Danh sách là một tập sắp thứ tự các phần tử cùng một kiểu. Đối với danh sách, người ta có một số thao tác: Tìm một phần tử trong danh sách, chèn một phần tử vào danh sách, xoá một phần tử khỏi danh sách, sắp xếp lại các phần tử trong danh sách theo một trật tự nào đó v.v...

4.2. BIỂU DIỄN DANH SÁCH TRONG MÁY TÍNH


Việc cài đặt một danh sách trong máy tính tức là tìm một cấu trúc dữ liệu cụ thể mà máy tính hiểu được để lưu các phần tử của danh sách đồng thời viết các đoạn chương trình con mô tả các thao tác cần thiết đối với danh sách.

4.2.1. Cài đặt bằng mảng một chiều

Khi cài đặt danh sách bằng một mảng, thì có một biến nguyên n lưu số phần tử hiện có trong danh sách. Nếu mảng được đánh số bắt đầu từ 1 thì các phần tử trong danh sách được cất giữ trong mảng bằng các phần tử được đánh số từ 1 tới n.


Chèn phần tử vào mảng:

Mảng ban đầu:




Nếu muốn chèn một phần tử V vào mảng tại vị trí p, ta phải:

- ❖ Dồn tất cả các phần tử từ vị trí p tới vị trí n về sau một vị trí:




- ❖ Đặt giá trị V vào vị trí p:



- ❖ Tăng n lên 1

Xoá phần tử khỏi mảng


Mảng ban đầu:




Muốn xoá phần tử thứ p của mảng mà vẫn giữ nguyên thứ tự các phần tử còn lại, ta phải:

- ❖ Dồn tất cả các phần tử từ vị trí p + 1 tới vị trí n lên trước một vị trí:

ĐHSPHN 1999-2004



- ❖ Giảm n đi 1




Trong trường hợp cần xóa một phần tử mà không cần duy trì thứ tự của các phần tử khác, ta chỉ cần đảo giá trị của phần tử cần xóa cho phần tử cuối cùng rồi giảm số phần tử của mảng (n) đi 1.

4.2.2. Cài đặt bằng danh sách nối đơn


Danh sách nối đơn gồm các nút được nối với nhau theo một chiều. Mỗi nút là một bản ghi (record) gồm hai trường:

- ❖ Trường thứ nhất chứa giá trị lưu trong nút đó
- ❖ Trường thứ hai chứa liên kết (con trỏ) tới nút kế tiếp, tức là chứa một thông tin đủ để biết nút kế tiếp nút đó trong danh sách là nút nào, trong trường hợp là nút cuối cùng (không có nút kế tiếp), trường liên kết này được gán một giá trị đặc biệt.



Hình 7: Cấu trúc nút của danh sách nối đơn


Nút đầu tiên trong danh sách được gọi là chốt của danh sách nối đơn (Head). Để duyệt danh sách nối đơn, ta bắt đầu từ chốt, dựa vào trường liên kết để đi sang nút kế tiếp, đến khi gặp giá trị đặc biệt (duyệt qua nút cuối) thì dừng lại



Hình 8: Danh sách nối đơn


Chèn phần tử vào danh sách nối đơn:

Danh sách ban đầu:




Muốn chèn thêm một nút chứa giá trị V vào vị trí của nút p, ta phải:

- ❖ Tạo ra một nút mới NewNode chứa giá trị V:




- ❖ Tìm nút q là nút đứng trước nút p trong danh sách (nút có liên kết tới p).
Nếu tìm thấy thì chỉnh lại liên kết: q liên kết tới NewNode, NewNode liên kết tới p



Nếu không có nút đứng trước nút p trong danh sách thì tức là p = Head, ta chỉnh lại liên kết: NewNode liên kết tới Head (cũ) và đặt lại Head = NewNode


Xoá phần tử khỏi danh sách nối đơn:

Danh sách ban đầu:



Muốn huỷ nút p khỏi danh sách nối đơn, ta phải:

- ❖ Tìm nút q là nút đứng liền trước nút p trong danh sách (nút có liên kết tới p)
Nếu tìm thấy thì chỉnh lại liên kết: q liên kết thẳng tới nút liền sau p, khi đó quá trình duyệt danh sách bắt đầu từ Head khi duyệt tới q sẽ nhảy qua không duyệt p nữa. Trên thực tế khi cài đặt bằng các biến động và con trỏ, ta nên có thao tác giải phóng bộ nhớ đã cấp cho nút p




Nếu không có nút đứng trước nút p trong danh sách thì tức là p = Head, ta chỉ việc đặt lại Head bằng nút đứng kế tiếp Head (cũ) trong danh sách. Sau đó có thể giải phóng bộ nhớ đã cấp cho nút p (Head cũ)

4.2.3. Cài đặt bằng danh sách nối kép


Danh sách nối kép gồm các nút được nối với nhau theo hai chiều. Mỗi nút là một bản ghi (record) gồm ba trường:

- ❖ Trường thứ nhất chứa giá trị lưu trong nút đó
- ❖ Trường thứ hai (Next) chứa liên kết (con trỏ) tới nút kế tiếp, tức là chứa một thông tin đủ để biết nút kế tiếp nút đó là nút nào, trong trường hợp là nút cuối cùng (không có nút kế tiếp), trường liên kết này được gán một giá trị đặc biệt.
- ❖ Trường thứ ba (Prev) chứa liên kết (con trỏ) tới nút liền trước, tức là chứa một thông tin đủ để biết nút đứng trước nút đó trong danh sách là nút nào, trong trường hợp là nút đầu tiên (không có nút liền trước) trường này được gán một giá trị đặc biệt.



Hình 9: Cấu trúc nút của danh sách nối kép

Khác với danh sách nối đơn, danh sách nối kép có hai chốt: Nút đầu tiên trong danh sách được gọi là First, nút cuối cùng trong danh sách được gọi là Last. Để duyệt danh sách nối kép, ta có hai cách: Hoặc bắt đầu từ First, dựa vào liên kết Next để đi sang nút kế tiếp, đến khi gặp giá trị đặc biệt (đi qua nút cuối) thì dừng lại. Hoặc bắt đầu từ Last, dựa vào liên kết Prev để đi sang nút liền trước, đến khi gặp giá trị đặc biệt (đi qua nút đầu) thì dừng lại




Hình 10: Danh sách nối kép

Việc chèn / xoá vào danh sách nối kép cũng đơn giản chỉ là kỹ thuật chỉnh lại các mối liên kết giữa các nút cho hợp lý, ta coi như bài tập.

4.2.4. Cài đặt bằng danh sách nối vòng một hướng

Trong danh sách nối đơn, phần tử cuối cùng trong danh sách có trường liên kết được gán một giá trị đặc biệt (thường sử dụng nhất là giá trị nil). Nếu ta cho trường liên kết của phần tử cuối cùng trở thẳng về phần tử đầu tiên của danh sách thì ta sẽ được một kiểu danh sách mới gọi là danh sách nối vòng một hướng.




Hình 11: Danh sách nối vòng một hướng

Đối với danh sách nối vòng, ta chỉ cần biết một nút bất kỳ của danh sách là ta có thể duyệt được hết các nút trong danh sách bằng cách đi theo hướng của các liên kết. Chính vì lý do này, khi chèn xoá vào danh sách nối vòng, ta không phải xử lý các trường hợp riêng khi chèn xoá tại vị trí của chốt

4.2.5. Cài đặt bằng danh sách nối vòng hai hướng

Danh sách nối vòng một hướng chỉ cho ta duyệt các nút của danh sách theo một chiều, nếu cài đặt bằng danh sách nối vòng hai hướng thì ta có thể duyệt các nút của danh sách cả theo chiều ngược lại nữa. Danh sách nối vòng hai hướng có thể tạo thành từ danh sách nối kép nếu ta cho trường Prev của nút First trở thẳng tới nút Last còn trường Next của nút Last thì trở thẳng về nút First.



Hình 12: Danh sách nối vòng hai hướng

Bài tập

Bài 1

Lập chương trình quản lý danh sách học sinh, tuỳ chọn loại danh sách cho phù hợp, chương trình có những chức năng sau: (Hồ sơ một học sinh giả sử có: Tên, lớp, số điện thoại, điểm TB ...)

Cho phép nhập danh sách học sinh từ bàn phím hay từ file.

Cho phép in ra danh sách học sinh gồm có tên và xếp loại

Cho phép in ra danh sách học sinh gồm các thông tin đầy đủ

Cho phép nhập vào từ bàn phím một tên học sinh và một tên lớp, tìm xem có học sinh có tên nhập vào trong lớp đó không ?. Nếu có thì in ra số điện thoại của học sinh đó

Cho phép vào một hồ sơ học sinh mới từ bàn phím, bổ sung học sinh đó vào danh sách học sinh, in ra danh sách mới.

Cho phép nhập vào từ bàn phím tên một lớp, loại bỏ tất cả các học sinh của lớp đó khỏi danh sách, in ra danh sách mới.

Có chức năng sắp xếp danh sách học sinh theo thứ tự giảm dần của điểm trung bình

Cho phép nhập vào hồ sơ một học sinh mới từ bàn phím, chèn học sinh đó vào danh sách mà không làm thay đổi thứ tự đã sắp xếp, in ra danh sách mới.

Cho phép lưu trữ lại trên đĩa danh sách học sinh khi đã thay đổi.

Bài 2

Có n người đánh số từ 1 tới n ngồi quanh một vòng tròn ($n \leq 10000$), cùng chơi một trò chơi: Một người nào đó đếm 1, người kế tiếp, theo chiều kim đồng hồ đếm 2... cứ như vậy cho tới người đếm đến một số nguyên tố thì phải ra khỏi vòng tròn, người kế tiếp lại đếm bắt đầu từ 1: Hãy lập chương trình

Nhập vào 2 số n và S từ bàn phím

- ❖ Cho biết nếu người thứ nhất là người đếm 1 thì người còn lại cuối cùng trong vòng tròn là người thứ mấy
- ❖ Cho biết nếu người còn lại cuối cùng trong vòng tròn là người thứ k thì người đếm 1 là người nào?

Giải quyết hai yêu cầu trên trong trường hợp: đầu tiên trò chơi được đếm theo chiều kim đồng hồ, khi có một người bị ra khỏi cuộc chơi thì vẫn là người kế tiếp đếm 1 nhưng quá trình đếm ngược lại (tức là ngược chiều kim đồng hồ)

§5. NGĂN XẾP VÀ HÀNG ĐỢI

5.1. NGĂN XẾP (STACK)

Ngăn xếp là một kiểu danh sách được trang bị hai phép toán **bổ sung một phần tử** vào cuối danh sách và **loại bỏ một phần tử** cũng ở cuối danh sách.

Có thể hình dung ngăn xếp như hình ảnh một chồng đĩa, đĩa nào được đặt vào chồng sau cùng sẽ nằm trên tất cả các đĩa khác và sẽ được lấy ra đầu tiên. Vì nguyên tắc “vào sau ra trước” đó, Stack còn có tên gọi là danh sách kiểu LIFO (Last In First Out) và vị trí cuối danh sách được gọi là đỉnh (Top) của Stack.

5.1.1. Mô tả Stack bằng mảng

Khi mô tả Stack bằng mảng:

- ❖ Việc bổ sung một phần tử vào Stack tương đương với việc thêm một phần tử vào cuối mảng.
- ❖ Việc loại bỏ một phần tử khỏi Stack tương đương với việc loại bỏ một phần tử ở cuối mảng.
- ❖ Stack bị tràn khi bổ sung vào mảng đã đầy
- ❖ Stack là rỗng khi số phần tử thực sự đang chứa trong mảng = 0.

```
program StackByArray;
const
  max = 10000;
var
  Stack: array[1..max] of Integer;
  Top: Integer; {Top lưu chỉ số phần tử cuối trong Stack}

procedure StackInit; {Khởi tạo Stack rỗng}
begin
  Top := 0;
end;

procedure Push(V: Integer); {Đẩy một giá trị V vào Stack}
begin
  if Top = max then writeln('Stack is full') {Nếu Stack đã đầy thì không đẩy được thêm vào nữa}
  else
    begin
      Inc(Top); Stack[Top] := V; {Nếu không thì thêm một phần tử vào cuối mảng}
    end;
end;

function Pop: Integer; {Lấy một giá trị ra khỏi Stack, trả về trong kết quả hàm}
begin
  if Top = 0 then writeln('Stack is empty') {Stack đang rỗng thì không lấy được}
  else
    begin
      Pop := Stack[Top]; Dec(Top); {Lấy phần tử cuối ra khỏi mảng}
    end;
end;

begin
  StackInit;
```

```

{Test}; {Đưa một vài lệnh để kiểm tra hoạt động của Stack}
end.

```

Khi cài đặt bằng mảng, tuy các thao tác đối với Stack viết hết sức đơn giản nhưng ở đây ta vẫn chia thành các chương trình con, mỗi chương trình con mô tả một thao tác, để từ đó về sau, ta chỉ cần biết rằng chương trình của ta có một cấu trúc Stack, còn ta mô phỏng cụ thể như thế nào thì không cần phải quan tâm nữa, và khi cài đặt Stack bằng các cấu trúc dữ liệu khác, chỉ cần sửa lại các thủ tục StackInit, Push và Pop mà thôi.

5.1.2. Mô tả Stack bằng danh sách nối đơn kiểu LIFO

Khi cài đặt Stack bằng danh sách nối đơn kiểu LIFO, thì Stack bị tràn khi vùng không gian nhớ dùng cho các biến động không còn đủ để thêm một phần tử mới. Tuy nhiên, việc kiểm tra điều này rất khó bởi nó phụ thuộc vào máy tính và ngôn ngữ lập trình. Ví dụ như đối với Turbo Pascal, khi Heap còn trống 80 Bytes thì cũng chỉ đủ chỗ cho 10 biến, mỗi biến 6 Bytes mà thôi. Mặt khác, không gian bộ nhớ dùng cho các biến động thường rất lớn nên cài đặt dưới đây ta bỏ qua việc kiểm tra Stack tràn.

```

program StackByLinkedList;
type
  PNode = ^TNode; {Con trỏ tới một nút của danh sách}
  TNode = record {Cấu trúc một nút của danh sách}
    Value: Integer;
    Link: PNode;
  end;
var
  Top: PNode; {Con trỏ định Stack}

procedure StackInit; {Khởi tạo Stack rỗng}
begin
  Top := nil;
end;

procedure Push(V: Integer); {Đẩy giá trị V vào Stack ⇔ thêm nút mới chứa V và nối nút đó vào danh sách}
var
  P: PNode;
begin
  New(P); P^.Value := V; {Tạo ra một nút mới}
  P^.Link := Top; Top := P; {Móc nút đó vào danh sách}
end;

function Pop: Integer; {Lấy một giá trị ra khỏi Stack, trả về trong kết quả hàm}
var
  P: PNode;
begin
  if Top = nil then writeln('Stack is empty')
  else
    begin
      Pop := Top^.Value; {Gán kết quả hàm}
      P := Top^.Link; {Giữ lại nút tiếp theo Top^ (nút được đẩy vào danh sách trước nút Top^)}
      Dispose(Top); Top := P; {Giải phóng bộ nhớ cấp cho Top^, cập nhật lại Top mới}
    end;
end;

begin
  StackInit;
  {Test}; {Đưa một vài lệnh để kiểm tra hoạt động của Stack}
end.

```

5.2. HÀNG ĐỢI (QUEUE)

Hàng đợi là một kiểu danh sách được trang bị hai phép toán **bổ sung một phần tử** vào cuối danh sách (Rear) và **loại bỏ một phần tử** ở đầu danh sách (Front).

Có thể hình dung hàng đợi như một đoàn người xếp hàng mua vé: Người nào xếp hàng trước sẽ được mua vé trước. Vì nguyên tắc “vào trước ra trước” đó, Queue còn có tên gọi là danh sách kiểu FIFO (First In First Out).

5.2.1. Mô tả Queue bằng mảng

Khi mô tả Queue bằng mảng, ta có hai chỉ số Front và Rear, Front lưu chỉ số phần tử đầu Queue còn Rear lưu chỉ số cuối Queue, khởi tạo Queue rỗng: Front := 1 và Rear := 0;

- ❖ Để thêm một phần tử vào Queue, ta tăng Rear lên 1 và đưa giá trị đó vào phần tử thứ Rear.
- ❖ Để loại một phần tử khỏi Queue, ta lấy giá trị ở vị trí Front và tăng Front lên 1.
- ❖ Khi Rear tăng lên hết khoảng chỉ số của mảng thì mảng đã đầy, không thể đầy thêm phần tử vào nữa.
- ❖ Khi Front > Rear thì tức là Queue đang rỗng

Như vậy chỉ một phần của mảng từ vị trí Front tới Rear được sử dụng làm Queue.

```
program QueueByArray;
const
  max = 10000;
var
  Queue: array[1..max] of Integer;
  Front, Rear: Integer;

procedure QueueInit; {Khởi tạo một hàng đợi rỗng}
begin
  Front := 1; Rear := 0;
end;

procedure Push(V: Integer); {Đẩy V vào hàng đợi}
begin
  if Rear = max then writeln('Overflow')
  else
    begin
      Inc(Rear);
      Queue[Rear] := V;
    end;
end;


function Pop: Integer; {Lấy một giá trị khỏi hàng đợi, trả về trong kết quả hàm}
begin
  if Front > Rear then writeln('Queue is Empty')
  else
    begin
      Pop := Queue[Front];
      Inc(Front);
    end;
end;

begin
  QueueInit;
  {Test}; {Đưa một vài lệnh để kiểm tra hoạt động của Queue}
end.
```

5.2.2. Mô tả Queue bằng danh sách vòng

Xem lại chương trình cài đặt Stack bằng một mảng kích thước tối đa 10000 phần tử, ta thấy rằng nếu như ta làm 6000 lần Push rồi 6000 lần Pop rồi lại 6000 lần Push thì vẫn không có vấn đề gì xảy ra. Lý do là vì chỉ số Top lưu đỉnh của Stack sẽ được tăng lên 6000 rồi lại giảm đến 0 rồi lại tăng trở lại lên 6000. Nhưng đối với cách cài đặt Queue như trên thì sẽ gặp thông báo lỗi tràn mảng, bởi mỗi lần Push, chỉ số cuối hàng đợi Rear cũng tăng lên và không bao giờ bị giảm đi cả. Đó chính là nhược điểm mà ta nói tới khi cài đặt: Chỉ có các phần tử từ vị trí Front tới Rear là thuộc Queue, các phần tử từ vị trí 1 tới Front - 1 là vô nghĩa.

Để khắc phục điều này, ta mô tả Queue bằng một danh sách vòng (biểu diễn bằng mảng hoặc cấu trúc liên kết), coi như các phần tử của mảng được xếp quanh vòng theo một hướng nào đó. Các phần tử nằm trên phần cung tròn từ vị trí Front tới vị trí Rear là các phần tử của Queue. Có thêm một biến n lưu số phần tử trong Queue. Việc thêm một phần tử vào Queue tương đương với việc dịch chỉ số Rear theo vòng một vị trí rồi đặt giá trị mới vào đó. Việc loại bỏ một phần tử trong Queue tương đương với việc lấy ra phần tử tại vị trí Front rồi dịch chỉ số Front theo vòng.



Hình 13: Dùng danh sách vòng mô tả Queue

Lưu ý là trong thao tác Push và Pop phải kiểm tra Queue tràn hay Queue cạn nên phải cập nhật lại biến n . (Ở đây dùng thêm biến n cho dễ hiểu còn trên thực tế chỉ cần hai biến Front và Rear là ta có thể kiểm tra được Queue tràn hay cạn rồi)

```

program QueueByCList;
const
  max = 10000;
var
  Queue: array[0..max - 1] of Integer;
  i, n, Front, Rear: Integer;

procedure QueueInit; {Khởi tạo Queue rỗng}
begin
  Front := 0; Rear := max - 1; n := 0;
end;

procedure Push(V: Integer); {Đẩy giá trị V vào Queue}
begin
  if n = max then writeln('Queue is Full')
  else
    begin
      Rear := (Rear + 1) mod max; {Rear chạy theo vòng tròn}
      Queue[Rear] := V;
    end;
  n := n + 1;
end;
  
```

```

    Inc(n);
  end;
end;

function Pop: Integer; {Lấy một phần tử khỏi Queue, trả về trong kết quả hàm}
begin
  if n = 0 then WriteLn('Queue is Empty')
  else
    begin
      Pop := Queue[Front];
      Front := (Front + 1) mod max; {Front chạy theo vòng tròn}
      Dec(n);
    end;
end;

begin
  QueueInit;
  {Test}; {Đưa một vài lệnh để kiểm tra hoạt động của Queue}
end.

```

5.2.3. Mô tả Queue bằng danh sách nối đơn kiểu FIFO

Tương tự như cài đặt Stack bằng danh sách nối đơn kiểu LIFO, ta cũng không kiểm tra Queue tràn trong trường hợp mô tả Queue bằng danh sách nối đơn kiểu FIFO.

```

program QueueByLinkedList;
type
  PNode = ^TNode; {Kiểu con trỏ tới một nút của danh sách}
  TNode = record {Cấu trúc một nút của danh sách}
    Value: Integer;
    Link: PNode;
  end;
var
  Front, Rear: PNode; {Hai con trỏ tới nút đầu và nút cuối của danh sách}

procedure QueueInit; {Khởi tạo Queue rỗng}
begin
  Front := nil;
end;

procedure Push(V: Integer); {Đẩy giá trị V vào Queue}
var
  P: PNode;
begin
  New(P); P^.Value := V; {Tạo ra một nút mới}
  P^.Link := nil;
  if Front = nil then Front := P {Mở nút đó vào danh sách}
  else Rear^.Link := P;
  Rear := P; {Nút mới trở thành nút cuối, cập nhật lại con trỏ Rear}
end;

function Pop: Integer; {Lấy giá trị khỏi Queue, trả về trong kết quả hàm}
var
  P: PNode;
begin
  if Front = nil then WriteLn('Queue is empty')
  else
    begin
      Pop := Front^.Value; {Gán kết quả hàm}
      P := Front^.Link; {Giữ lại nút tiếp theo Front^ (Nút được đẩy vào danh sách ngay sau Front^)}
      Dispose(Front); Front := P; {Giải phóng bộ nhớ cấp cho Front^, cập nhật lại Front mới}
    end;
end;

```

```

begin
    QueueInit;
    {Test}; {Đưa một vài lệnh để kiểm tra hoạt động của Queue}
end.

```

Bài tập


Bài 1

Tìm hiểu cơ chế xếp chồng của thủ tục đệ quy, phương pháp dùng ngăn xếp để khử đệ quy.

Viết chương trình mô tả cách đổi cơ số từ hệ thập phân sang hệ cơ số R dùng ngăn xếp

Bài 2

Hình 14 là cơ cấu đường tàu tại một ga xe lửa



Hình 14: Di chuyển toa tàu

Ban đầu ở đường ray A chứa các toa tàu đánh số từ 1 tới n theo thứ tự từ trái qua phải, người ta muốn chuyển các toa đó sang đường ray C để được một thứ tự mới là một hoán vị của (1, 2, ..., n) theo quy tắc: chỉ được đưa các toa tàu chạy theo đường ray theo hướng mũi tên, có thể dùng đoạn đường ray B để chứa tạm các toa tàu trong quá trình di chuyển.

a) Hãy nhập vào hoán vị cần có, cho biết có phương án chuyển hay không, và nếu có hãy đưa ra cách chuyển:


Ví dụ: n = 4; Thứ tự cần có (1, 4, 3, 2)

1) A → C; 2) A → B; 3) A → B; 4) A → C; 5) B → C; 6) B → C

b) Những hoán vị nào của thứ tự các toa là có thể tạo thành trên đoạn đường ray C với luật di chuyển như trên

Bài 3

Tương tự như bài trên, nhưng với sơ đồ đường ray sau:



Hình 15: Di chuyển toa tàu (2)

§6. CÂY (TREE)

6.1. ĐỊNH NGHĨA


Cấu trúc dữ liệu trùu tượng ta quan tâm tới trong mục này là cấu trúc cây. Cây là một cấu trúc dữ liệu gồm một tập hữu hạn các nút, giữa các nút có một quan hệ phân cấp gọi là quan hệ “cha – con”. Có một nút đặc biệt gọi là gốc (root).

Có thể định nghĩa cây bằng các đệ quy như sau:

- ❖ Mỗi nút là một cây, nút đó cũng là gốc của cây ấy
- ❖ Nếu n là một nút và n_1, n_2, \dots, n_k lần lượt là gốc của các cây T_1, T_2, \dots, T_k ; các cây này đôi một không có nút chung. Thì nếu cho nút n trở thành cha của các nút n_1, n_2, \dots, n_k ta sẽ được một cây mới T . Cây này có nút n là gốc còn các cây T_1, T_2, \dots, T_k trở thành các cây con (subtree) của gốc.

Để tiện, người ta còn cho phép tồn tại một cây không có nút nào mà ta gọi là cây rỗng (null tree).

Xét cây trong Hình 16:



Hình 16: Cây


A là cha của B, C, D, còn G, H, I là con của D

Số các con của một nút được gọi là **cấp của nút** đó, ví dụ cấp của A là 3, cấp của B là 2, cấp của C là 0.

Nút có cấp bằng 0 được gọi là **nút lá** (leaf) hay nút tận cùng. Ví dụ như ở trên, các nút E, F, C, G, J, K và I là các nút lá. Những nút không phải là lá được gọi là **nút nhánh** (branch)

Cấp cao nhất của một nút trên cây gọi là **cấp của cây** đó, cây ở hình trên là cây cấp 3.

Gốc của cây người ta gán cho số mức là 1, nếu nút cha có mức là i thì nút con sẽ có mức là $i + 1$. Mức của cây trong Hình 16 được chỉ ra trong Hình 17:


**Hình 17: Mức của các nút trên cây**

Chiều cao (height) hay **chiều sâu** (depth) của một cây là số mức lớn nhất của nút có trên cây đó. Cây ở trên có chiều cao là 4

Một tập hợp các cây phân biệt được gọi là **rừng** (forest), một cây cũng là một rừng. Nếu bỏ nút gốc trên cây thì sẽ tạo thành một rừng các cây con.

Ví dụ:

- ❖ Mục lục của một cuốn sách với phần, chương, bài, mục v.v... có cấu trúc của cây
- ❖ Cấu trúc thư mục trên đĩa cũng có cấu trúc cây, thư mục gốc có thể coi là gốc của cây đó với các cây con là các thư mục con và tệp nằm trên thư mục gốc.
- ❖ Gia phả của một họ tộc cũng có cấu trúc cây.
- ❖ Một biểu thức số học gồm các phép toán cộng, trừ, nhân, chia cũng có thể lưu trữ trong một cây mà các toán hạng được lưu trữ ở các nút lá, các toán tử được lưu trữ ở các nút nhánh, mỗi nhánh là một biểu thức con.


**Hình 18: Cây biểu diễn biểu thức**

6.2. CÂY NHỊ PHÂN (BINARY TREE)

Cây nhị phân là một dạng quan trọng của cấu trúc cây. Nó có đặc điểm là mọi nút trên cây chỉ có tối đa hai nhánh con. Với một nút thì người ta cũng phân biệt cây con trái và cây con phải của nút đó. Cây nhị phân là cây có tính đến thứ tự của các nhánh con.


Cần chú ý tới một số dạng đặc biệt của cây nhị phân

Các cây nhị phân trong Hình 19 được gọi là **cây nhị phân suy biến** (degenerate binary tree), các nút không phải lá chỉ có một nhánh con. Cây a) được gọi là cây lệch phải, cây b) được gọi là cây lệch trái, cây c) và d) được gọi là cây zíc-zắc.



Hình 19: Các dạng cây nhị phân suy biến

Các cây trong Hình 20 được gọi là **cây nhị phân hoàn chỉnh** (complete binary tree): Nếu chiều cao của cây là h thì mọi nút có mức $< h - 1$ đều có đúng 2 nút con. Còn nếu mọi nút có mức $\leq h - 1$ đều có đúng 2 nút con như trường hợp cây f) ở trên thì cây đó được gọi là **cây nhị phân đầy đủ** (full binary tree). Cây nhị phân đầy đủ là trường hợp riêng của cây nhị phân hoàn chỉnh.



Hình 20: Cây nhị phân hoàn chỉnh và cây nhị phân đầy đủ


Ta có thể thấy ngay những tính chất sau bằng phép chứng minh quy nạp:

- ❖ Trong các cây nhị phân có cùng số lượng nút như nhau thì cây nhị phân suy biến có chiều cao lớn nhất, còn cây nhị phân hoàn chỉnh thì có chiều cao nhỏ nhất.
- ❖ Số lượng tối đa các nút trên mức i của cây nhị phân là 2^{i-1} , tối thiểu là 1 ($i \geq 1$).
- ❖ Số lượng tối đa các nút trên một cây nhị phân có chiều cao h là $2^h - 1$, tối thiểu là h ($h \geq 1$).
- ❖ Cây nhị phân hoàn chỉnh có n nút thì chiều cao của nó là $h = \lfloor \lg(n) \rfloor + 1$.

6.3. BIỂU ĐIỂN CÂY NHỊ PHÂN

6.3.1. Biểu diễn bằng mảng

Nếu có một cây nhị phân đầy đủ, ta có thể dễ dàng đánh số cho các nút trên cây đó theo thứ tự lần lượt từ mức 1 trở đi, hết mức này đến mức khác và từ trái sang phải đối với các nút ở mỗi mức.



Hình 21: Đánh số các nút của cây nhị phân đầy đủ để biểu diễn bằng mảng


Với cách đánh số này, con của nút thứ i sẽ là các nút thứ $2i$ và $2i + 1$. Cha của nút thứ j là nút $j \div 2$. Từ đó có thể lưu trữ cây bằng một mảng T , nút thứ i của cây được lưu trữ bằng phần tử $T[i]$.

Với cây nhị phân đầy đủ ở Hình 21 thì khi lưu trữ bằng mảng, ta sẽ được mảng như sau:

1	2	3	4	5	6	7
A	B	E	C	D	F	G

Trong trường hợp cây nhị phân không đầy đủ, ta có thể thêm vào một số nút giả để được cây nhị phân đầy đủ, và gán những giá trị đặc biệt cho những phần tử trong mảng T tương ứng với những nút này. Hoặc dùng thêm một mảng phụ để đánh dấu những nút nào là nút giả ta thêm vào. Chính vì lý do này nên với cây nhị phân không đầy đủ, ta sẽ gặp phải sự lãng phí bộ nhớ vì có thể sẽ phải thêm rất nhiều nút giả vào thì mới được cây nhị phân đầy đủ.

Ví dụ với cây lệch trái, ta phải dùng một mảng 31 phần tử để lưu cây nhị phân chỉ gồm 5 nút



Hình 22: Nhược điểm của phương pháp biểu diễn cây bằng mảng

6.3.2. Biểu diễn bằng cấu trúc liên kết.


Khi biểu diễn cây nhị phân bằng cấu trúc liên kết, mỗi nút của cây là một bản ghi (record) gồm 3 trường:

- ❖ Trường Info: Chứa giá trị lưu tại nút đó
- ❖ Trường Left: Chứa liên kết (con trỏ) tới nút con trái, tức là chứa một thông tin đủ để biết nút con trái của nút đó là nút nào, trong trường hợp không có nút con trái, trường này được gán một giá trị đặc biệt.
- ❖ Trường Right: Chứa liên kết (con trỏ) tới nút con phải, tức là chứa một thông tin đủ để biết nút con phải của nút đó là nút nào, trong trường hợp không có nút con phải, trường này được gán một giá trị đặc biệt.



Hình 23: Cấu trúc nút của cây nhị phân

Đối với cây ta chỉ cần phải quan tâm giữ lại nút gốc, bởi từ nút gốc, đi theo các hướng liên kết Left, Right ta có thể duyệt mọi nút khác.



Hình 24: Biểu diễn cây bằng cấu trúc liên kết

6.4. PHÉP DUYỆT CÂY NHỊ PHÂN

Phép xử lý các nút trên cây mà ta gọi chung là phép thăm (Visit) các nút một cách hệ thống sao cho mỗi nút chỉ được thăm một lần gọi là phép duyệt cây.

Giả sử rằng nếu như một nút không có nút con trái (hoặc nút con phải) thì liên kết Left (Right) của nút đó được liên kết thẳng tới một nút đặc biệt mà ta gọi là NIL (hay NULL), nếu cây rỗng thì nút gốc của cây đó cũng được gán bằng NIL. Khi đó có ba cách duyệt cây hay được sử dụng:

6.4.1. Duyệt theo thứ tự trước (preorder traversal)

Trong phép duyệt theo thứ tự trước thì giá trị trong mỗi nút bất kỳ sẽ được liệt kê trước giá trị lưu trong hai nút con của nó, có thể mô tả bằng thủ tục đệ quy sau:

```
procedure Visit(N); {Duyệt nhánh cây nhận N là nút gốc của nhánh đó}
begin
    if N ≠ nil then
        begin
            <Output trường Info của nút N>
            Visit(Nút con trái của N);
            Visit(Nút con phải của N);
        end;
    end;
```

Quá trình duyệt theo thứ tự trước bắt đầu bằng lời gọi Visit(nút gốc).

Như cây ở Hình 24, nếu ta duyệt theo thứ tự trước thì các giá trị sẽ lần lượt được liệt kê theo thứ tự:

A B D H I E J C F K G L

6.4.2. Duyệt theo thứ tự giữa (inorder traversal)

Trong phép duyệt theo thứ tự giữa thì giá trị trong mỗi nút bất kỳ sẽ được liệt kê sau giá trị lưu ở nút con trái và được liệt kê trước giá trị lưu ở nút con phải của nút đó, có thể mô tả bằng thủ tục đệ quy sau:

```
procedure Visit(N); {Duyệt nhánh cây nhận N là nút gốc của nhánh đó}
begin
    if N ≠ nil then
        begin
            Visit(Nút con trái của N);
            <Output trường Info của nút N>
            Visit(Nút con phải của N);
        end;
    end;
```

Quá trình duyệt theo thứ tự giữa cũng bắt đầu bằng lời gọi Visit(nút gốc).

Như cây ở Hình 24, nếu ta duyệt theo thứ tự giữa thì các giá trị sẽ lần lượt được liệt kê theo thứ tự:

H D I B E J A K F C G L

6.4.3. Duyệt theo thứ tự sau (postorder traversal)

Trong phép duyệt theo thứ tự sau thì giá trị trong mỗi nút bất kỳ sẽ được liệt kê sau giá trị lưu ở hai nút con của nút đó, có thể mô tả bằng thủ tục đệ quy sau:

```
procedure Visit(N); {Duyệt nhánh cây nhận N là nút gốc của nhánh đó}
begin
    if N ≠ nil then
        begin
            Visit(Nút con trái của N);
            Visit(Nút con phải của N);
            <Output trường Info của nút N>
        end;
    end;
```

Quá trình duyệt theo thứ tự sau cũng bắt đầu bằng lời gọi Visit(nút gốc).

Cũng với cây ở Hình 24, nếu ta duyệt theo thứ tự sau thì các giá trị sẽ lần lượt được liệt kê theo thứ tự:

H I D J E B K F L G C A


6.5. CÂY K_PHÂN

Cây K_phân là một dạng cấu trúc cây mà mỗi nút trên cây có tối đa K nút con (có tính đến thứ tự của các nút con).

6.5.1. Biểu diễn cây K_phân bằng mảng

Cũng tương tự như việc biểu diễn cây nhị phân, người ta có thể thêm vào cây K_phân một số nút giả để cho mỗi nút nhánh của cây K_phân đều có đúng K nút con, các nút con được xếp thứ tự từ nút con thứ nhất tới nút con thứ K, sau đó đánh số các nút trên cây K_phân bắt đầu từ 0 trở đi, bắt đầu từ mức 1, hết mức này đến mức khác và từ “trái qua phải” ở mỗi mức.

Theo cách đánh số này, nút con thứ j của nút i là: $i * K + j$. Nút cha của nút x là nút $(x - 1) / K$. Ta có thể dùng một mảng T đánh số từ 0 để lưu các giá trị trên các nút: Giá trị tại nút thứ i được lưu trữ ở phần tử $T[i]$.



Hình 25: Đánh số các nút của cây 3_phân để biểu diễn bằng mảng

6.5.2. Biểu diễn cây K_phân bằng cấu trúc liên kết

Khi biểu diễn cây K_phân bằng cấu trúc liên kết, mỗi nút của cây là một bản ghi (record) gồm hai trường:

- ❖ Trường Info: Chứa giá trị lưu trong nút đó.
- ❖ Trường Links: Là một mảng gồm K phần tử, phần tử thứ i chứa liên kết (con trỏ) tới nút con thứ i, trong trường hợp không có nút con thứ i thì $Links[i]$ được gán một giá trị đặc biệt.

Đối với cây K_phân, ta cũng chỉ cần giữ lại nút gốc, bởi từ nút gốc, đi theo các hướng liên kết có thể đi tới mọi nút khác.

6.6. CÂY TỔNG QUÁT


Trong thực tế, có một số ứng dụng đòi hỏi một cấu trúc dữ liệu dạng cây nhưng không có ràng buộc gì về số con của một nút trên cây, ví dụ như cấu trúc thư mục trên đĩa hay hệ thống đề mục của một cuốn sách. Khi đó, ta phải tìm cách mô tả một cách khoa học cấu trúc dữ liệu dạng cây tổng quát. Cũng như trường hợp cây nhị phân, người ta thường biểu diễn cây tổng quát bằng hai cách: Lưu trữ kế tiếp bằng mảng và lưu trữ bằng cấu trúc liên kết.

6.6.1. Biểu diễn cây tổng quát bằng mảng

Để lưu trữ cây tổng quát bằng mảng, trước hết, ta đánh số các nút trên cây bắt đầu từ 1 theo một thứ tự tùy ý. Giả sử cây có n nút thì ta sử dụng:

- ❖ Một mảng Info[1..n], trong đó Info[i] là giá trị lưu trong nút thứ i.
- ❖ Một mảng Children được chia làm n đoạn, đoạn thứ i gồm một dãy liên tiếp các phần tử là chỉ số các nút con của nút i. Như vậy mảng Children sẽ chứa tất cả chỉ số của mọi nút con trên cây (ngoại trừ nút gốc) nên nó sẽ gồm $n - 1$ phần tử, lưu ý rằng khi chia mảng Children làm n đoạn thì sẽ có những đoạn rỗng (tương ứng với danh sách các nút con của một nút lá).
- ❖ Một mảng Head[1..n + 1], để đánh dấu vị trí cắt đoạn trong mảng Children: Head[i] là vị trí đứng liền trước đoạn thứ i, hay nói cách khác: Đoạn con tính từ chỉ số Head[i] + 1 đến Head[i] của mảng Children chứa chỉ số các nút con của nút thứ i. Khi Head[i] = Head[i+1] có nghĩa là đoạn thứ i rỗng. Quy ước: Head[n+1] = n - 1.
- ❖ Một biến lưu chỉ số của nút gốc.

Ví dụ:




Hình 26: Biểu diễn cây tổng quát bằng mảng

6.6.2. Lưu trữ cây tổng quát bằng cấu trúc liên kết

Khi lưu trữ cây tổng quát bằng cấu trúc liên kết, mỗi nút là một bản ghi (record) gồm ba trường:

- ❖ Trường Info: Chứa giá trị lưu trong nút đó.
- ❖ Trường FirstChild: Chứa liên kết (con trỏ) tới nút con đầu tiên của nút đó (con cả), trong trường hợp là nút lá (không có nút con), trường này được gán một giá trị đặc biệt.
- ❖ Trường Sibling: Chứa liên kết (con trỏ) tới nút em kế cận bên phải (nút cùng cha với nút đang xét, khi sắp thứ tự các con thì nút đó đứng liền sau nút đang xét). Trong trường hợp không có nút em kế cận bên phải, trường này được gán một giá trị đặc biệt.



Hình 27: Cấu trúc nút của cây tổng quát

Dễ thấy được tính đúng đắn của phương pháp biểu diễn, bởi từ một nút N bất kỳ, ta có thể đi theo liên kết FirstChild để đến nút con cả, nút này chính là chốt của một danh sách nối đơn các nút con của nút N: từ nút con cả, đi theo liên kết Sibling, ta có thể duyệt tất cả các nút con của nút N.

Bài tập

Bài 1

Viết chương trình mô tả cây nhị phân dùng cấu trúc liên kết, mỗi nút chứa một số nguyên, và viết các thủ tục duyệt trước, giữa, sau.

Bài 2

Chứng minh rằng nếu cây nhị phân có x nút lá và y nút cấp 2 thì $x = y + 1$

Bài 3

Chứng minh rằng nếu ta biết dãy các nút được thăm của một cây nhị phân khi duyệt theo thứ tự trước và thứ tự giữa thì có thể dựng được cây nhị phân đó. Điều này còn đúng nữa không đối với thứ tự trước và thứ tự sau? Với thứ tự giữa và thứ tự sau.

Bài 4


Viết các thủ tục duyệt trước, giữa, sau không đệ quy.

§7. KÝ PHÁP TIỀN TỐ, TRUNG TỐ VÀ HẬU TỐ

7.1. BIỂU THỨC DƯỚI DẠNG CÂY NHỊ PHÂN

Chúng ta có thể biểu diễn các biểu thức số học gồm các phép toán cộng, trừ, nhân, chia bằng một cây nhị phân, trong đó các nút lá biểu thị các hằng hay các biến (các toán hạng), các nút không phải là lá biểu thị các toán tử (phép toán số học chẵng hạn). Mỗi phép toán trong một nút sẽ tác động lên hai biểu thức con nằm ở cây con bên trái và cây con bên phải của nút đó.

Ví dụ: Cây biểu diễn biểu thức $(6 / 2 + 3) * (7 - 4)$



Hình 28: Biểu thức dưới dạng cây nhị phân

7.2. CÁC KÝ PHÁP CHO CÙNG MỘT BIỂU THỨC

Với cây nhị phân biểu diễn biểu thức trong Hình 28,

- ❖ Nếu duyệt theo thứ tự trước, ta sẽ được $* + / 6 2 3 - 7 4$, đây là **dạng tiền tố** (prefix) của biểu thức. Trong ký pháp này, toán tử được viết trước hai toán hạng tương ứng, người ta còn gọi ký pháp này là ký pháp Ba lan.
- ❖ Nếu duyệt theo thứ tự giữa, ta sẽ được $6 / 2 + 3 * 7 - 4$. Ký pháp này hơi mập mờ vì thiếu dấu ngoặc. Nếu thêm vào thủ tục duyệt inorder việc bổ sung các cặp dấu ngoặc vào mỗi biểu thức con sẽ thu được biểu thức $((6 / 2) + 3) * (7 - 4)$. Ký pháp này gọi là **dạng trung tố** (infix) của một biểu thức (Thực ra chỉ cần thêm các dấu ngoặc đủ để tránh sự mập mờ mà thôi, không nhất thiết phải thêm vào đầy đủ các cặp dấu ngoặc).
- ❖ Nếu duyệt theo thứ tự sau, ta sẽ được $6 2 / 3 + 7 4 - *$, đây là **dạng hậu tố** (postfix) của biểu thức. Trong ký pháp này toán tử được viết sau hai toán hạng, người ta còn gọi ký pháp này là ký pháp nghịch đảo Balan (Reverse Polish Notation - RPN)

Chỉ có dạng trung tố mới cần có dấu ngoặc, dạng tiền tố và hậu tố không cần phải có dấu ngoặc.

7.3. CÁCH TÍNH GIÁ TRỊ BIỂU THỨC

Có một vấn đề cần lưu ý là khi máy tính giá trị một biểu thức số học gồm các toán tử hai ngôi (toán tử gồm hai toán hạng như $+, -, *, /$) thì máy chỉ thực hiện được phép toán đó với hai toán

hạng. Nếu biểu thức phức tạp thì máy phải chia nhỏ và tính riêng từng biểu thức trung gian, sau đó mới lấy giá trị tìm được để tính tiếp. Ví dụ như biểu thức $1 + 2 + 4$ máy sẽ phải tính $1 + 2$ trước được kết quả là 3 sau đó mới đem 3 cộng với 4 chứ không thể thực hiện phép cộng một lúc ba số được.

Khi lưu trữ biểu thức dưới dạng cây nhị phân thì ta có thể coi **mỗi nhánh con của cây đó mô tả một biểu thức trung gian** mà máy cần tính khi xử lý biểu thức lớn. Như ví dụ trên, máy sẽ phải tính hai biểu thức $6 / 2 + 3$ và $7 - 4$ trước khi làm phép tính nhân cuối cùng. Để tính biểu thức $6 / 2 + 3$ thì máy lại phải tính biểu thức $6 / 2$ trước khi đem cộng với 3 .

Vậy để tính một biểu thức lưu trữ trong một nhánh cây nhị phân gốc ở nút n , máy sẽ tính gần giống như hàm đệ quy sau:

```
function Calculate(n): Value; {Tính biểu thức con trong nhánh cây gốc n}
begin
  if <Nút n chưa phải là một toán tử> then
    Calculate := <Giá trị chưa trong nút n>
  else {Nút n chứa một toán tử R}
    begin
      x := Calculate(nút con trái của n);
      y := Calculate(nút con phải của n);
      Calculate := x R y;
    end;
  end.
```

(Trong trường hợp lập trình trên các hệ thống song song, việc tính giá trị biểu thức ở cây con trái và cây con phải có thể tiến hành đồng thời làm giảm đáng kể thời gian tính toán biểu thức).

Để ý rằng khi tính toán biểu thức, máy sẽ phải quan tâm tới việc tính biểu thức ở hai nhánh con trước, rồi mới xét đến toán tử ở nút gốc. Điều đó làm ta nghĩ tới phép cây theo thứ tự sau và ký pháp hậu tố. Trong những năm đầu 1950, nhà lô-gic học người Ba Lan Jan Lukasiewicz đã chứng minh rằng biểu thức hậu tố không cần phải có dấu ngoặc vẫn có thể tính được một cách đúng đắn bằng cách **đọc lần lượt biểu thức từ trái qua phải** và dùng một Stack để lưu các kết quả trung gian:

Bước 1: Khởi tạo một Stack rỗng

Bước 2: Đọc lần lượt các phần tử của biểu thức RPN từ trái qua phải (phần tử này có thể là hằng, biến hay toán tử) với mỗi phần tử đó, ta kiểm tra:

- ❖ Nếu phần tử này là một toán hạng thì đẩy giá trị của nó vào Stack.
- ❖ Nếu phần tử này là một toán tử \circ , ta lấy từ Stack ra hai giá trị (y và x) sau đó áp dụng toán tử \circ đó vào hai giá trị vừa lấy ra, đẩy kết quả tìm được ($x \circ y$) vào Stack (ra hai vào một).

Bước 3: Sau khi kết thúc bước 2 thì toàn bộ biểu thức đã được đọc xong, trong Stack chỉ còn duy nhất một phần tử, phần tử đó chính là giá trị của biểu thức.

Ví dụ: Tính biểu thức $10 \ 2 / 3 + 7 \ 4 - *$ tương ứng với biểu thức trung tố $(10 / 2 + 3) * (7 - 4)$

Đọc	Xử lý	Stack
10	Đẩy vào Stack	10

Đọc	Xử lý	Stack
2	Đẩy vào Stack	10, 2
/	Lấy 2 và 10 khỏi Stack, Tính được $10 / 2 = 5$, đẩy 5 vào Stack	5
3	Đẩy vào Stack	5, 3
+	Lấy 3 và 5 khỏi Stack, tính được $5 + 3 = 8$, đẩy 8 vào Stack	8
7	Đẩy vào Stack	8, 7
4	Đẩy vào Stack	8, 7, 4
-	Lấy 4 và 7 khỏi Stack, tính được $7 - 4 = 3$, đẩy 3 vào Stack	8, 3
*	Lấy 3 và 8 khỏi Stack, tính được $8 * 3 = 24$, đẩy 24 vào Stack	24

Ta được kết quả là 24

Dưới đây ta sẽ viết một chương trình đơn giản tính giá trị biểu thức RPN.

- ❖ **Input:** File văn bản CALRPN.INP chỉ gồm 1 dòng có không quá 255 ký tự, chứa các số thực và các toán tử {+, -, *, /}. Quy định khuôn dạng bắt buộc là hai số liền nhau trong biểu thức RPN phải viết cách nhau ít nhất một dấu cách.
- ❖ **Output:** Kết quả biểu thức đó.

CALRPN.INP	CALRPN.OUT
10 2/3 + 7 4 -*	10 2 / 3 + 7 4 - * = 24.0000

Để quá trình đọc một phần tử trong biểu thức RPN được dễ dàng hơn, sau bước nhập liệu, ta có thể hiệu chỉnh đôi chút biểu thức RPN về khuôn dạng dễ đọc nhất. Chẳng hạn như thêm và bớt một số dấu cách trong Input để mỗi phần tử (toán hạng, toán tử) đều cách nhau đúng một dấu cách, thêm một dấu cách vào cuối biểu thức RPN. Khi đó quá trình đọc lần lượt các phần tử trong biểu thức RPN có thể làm như sau:

```

T := '';
for p := 1 to Length(RPN) do {Xét các ký tự trong biểu thức RPN từ trái qua phải}
  if RPN[p] ≠ ' ' then T := T + RPN[p] {Nếu RPN[p] không phải dấu cách thì nối ký tự đó vào T}
  else {Nếu RPN[p] là dấu cách thì phần tử đang đọc đã đọc xong, tiếp theo sẽ là phần tử khác}
    begin
      {Xử lý phần tử T};
      T := ''; {Chuẩn bị đọc phần tử mới}
    end;
  
```

Để đơn giản, chương trình không kiểm tra lỗi viết sai biểu thức RPN, việc đó chỉ là thao tác tóm tắt không phức tạp lắm, chỉ cần xem lại thuật toán và cài thêm các mô-đun bắt lỗi tại mỗi bước.

```

P_2_07_1.PAS * Tính giá trị biểu thức RPN
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program CalculateRPNEexpression;
const
  InputFile = 'CALRPN.INP';
  OutputFile = 'CALRPN.OUT';
  Opt = ['+', '-', '*', '/'];
var
  T, RPN: String;
  Stack: array[1..255] of Extended;

```

```

p, Top: Integer;
f: Text;
{Các thao tác đối với Stack}
procedure StackInit;
begin
  Top := 0;
end;

procedure Push(V: Extended);
begin
  Inc(Top); Stack[Top] := V;
end;

function Pop: Extended;
begin
  Pop := Stack[Top]; Dec(Top);
end;

procedure Refine(var S: String); {Hiệu chỉnh biểu thức RPN về khuôn dạng dễ đọc nhất}
var
  i: Integer;
begin
  S := S + ' ';
  for i := Length(S) - 1 downto 1 do {Thêm những dấu cách giữa toán hạng và toán tử}
    if (S[i] in Opt) or (S[i + 1] in Opt) then
      Insert(' ', S, i + 1);
  for i := Length(S) - 1 downto 1 do {Xoá những dấu cách thừa}
    if (S[i] = ' ') and (S[i + 1] = ' ') then Delete(S, i + 1, 1);
end;

procedure Process(T: String); {Xử lý phần tử T đọc được từ biểu thức RPN}
var
  x, y: Extended;
  e: Integer;
begin
  if not (T[1] in Opt) then {T là toán hạng}
    begin
      Val(T, x, e); Push(x); {Đổi T thành số và đẩy giá trị đó vào Stack}
    end
  else {T là toán tử}
    begin
      y := Pop; x := Pop; {Ra hai}
      case T[1] of
        '+': x := x + y;
        '-': x := x - y;
        '*': x := x * y;
        '/': x := x / y;
      end;
      Push(x); {Vào một}
    end;
end;

begin
  Assign(f, InputFile); Reset(f);
  Readln(f, RPN);
  Close(f);
  Refine(RPN);
  StackInit;
  T := '';
  for p := 1 to Length(RPN) do {Xét các ký tự của biểu thức RPN từ trái qua phải}
    if RPN[p] <> ' ' then T := T + RPN[p] {nếu không phải dấu cách thì nối nó vào sau xâu T}
    else {Nếu gặp dấu cách}
      begin

```

```

Process(T); {Xử lý phần tử vừa đọc xong}
T := ''; {Đặt lại T để chuẩn bị đọc phần tử mới}
end;
Assign(f, OutputFile); Rewrite(f);
Writeln(f, RPN, ' = ', Pop:0:4); {In giá trị biểu thức RPN được lưu trong Stack}
Close(f);
end.

```

7.4. CHUYỂN TỪ DẠNG TRUNG TỐ SANG DẠNG HẬU TỐ

Có thể nói rằng việc tính toán biểu thức viết bằng ký pháp nghịch đảo Balan là khoa học hơn, máy móc, và đơn giản hơn việc tính toán biểu thức viết bằng ký pháp trung tố. Chỉ riêng việc không phải xử lý dấu ngoặc đã cho ta thấy ưu điểm của ký pháp RPN. Chính vì lý do này, các chương trình dịch vẫn cho phép lập trình viên viết biểu thức trên ký pháp trung tố theo thói quen, nhưng trước khi dịch ra các lệnh máy thì tất cả các biểu thức đều được chuyển về dạng RPN. Vấn đề đặt ra là phải có một thuật toán chuyển biểu thức dưới dạng trung tố về dạng RPN một cách hiệu quả, và dưới đây ta trình bày thuật toán đó:

Thuật toán sử dụng một Stack để chứa các toán tử và dấu ngoặc mở. Thủ tục Push(V) để đẩy một phần tử vào Stack, hàm Pop để lấy ra một phần tử từ Stack, hàm Get để đọc giá trị phần tử nằm ở đỉnh Stack mà không lấy phần tử đó ra. Ngoài ra mức độ ưu tiên của các toán tử được quy định bằng hàm Priority như sau: Ưu tiên cao nhất là dấu "*" và "/" với Priority là 2, tiếp theo là dấu "+" và "-" với Priority là 1, ưu tiên thấp nhất là dấu ngoặc mở "(" với Priority là 0.

```

Stack := Ø;
for <Phần tử T đọc được từ biểu thức infix> do
{T có thể là hằng, biến, toán tử hoặc dấu ngoặc được đọc từ biểu thức infix theo thứ tự từ trái qua phải}
  case T of
    '(': Push(T);
    ')':
      repeat
        x := Pop;
        if x ≠ '(' then Output(x);
        until x = '(';
    '+', '-', '*', '/':
      begin
        while (Stack ≠ Ø) and (Priority(T) ≤ Priority(Get)) do Output(Pop);
        Push(T);
      end;
    else Output(T);
  end;
while (Stack ≠ Ø) do Output(Pop);

```

Ví dụ với biểu thức trung tố $(10 / 2 + 3) * (7 - 4)$

Đọc	Xử lý	Stack	Output
(Đẩy vào Stack	(
10	Output: "10"	(10
/	Phép "/" được ưu tiên hơn "(" ở đỉnh Stack, đẩy "/" vào Stack	(/	
2	Output: "2"	(/ 2	

Đọc	Xử lý	Stack	Output
+	Phép “+” ưu tiên không cao hơn “/” ở đỉnh Stack. Lấy “/” khỏi Stack, Output: “/” So sánh tiếp: Phép “+” ưu tiên cao hơn “(” ở đỉnh Stack, đẩy “+” vào Stack	(+	/
3	Output: 3	(+	3
)	Lấy ra và hiển thị các phần tử trong Stack tới khi lấy phải dấu “(”	∅	+
*	Stack đang là rỗng, đẩy * vào Stack	*	
(Đẩy vào Stack	*(
7	Output: “7”	*(7
-	Phép “-” ưu tiên hơn “(” ở đỉnh Stack, đẩy “-” vào Stack	*(-	
4	Output: “4”	*(-	4
)	Lấy ra và hiển thị các phần tử trong Stack tới khi lấy phải dấu “(”	*	-
Hết	Lấy ra và hiển thị hết các phần tử còn lại trong Stack		*

Dưới đây là chương trình chuyển biểu thức viết ở dạng trung tố sang dạng RPN. Biểu thức trung tố đầu vào sẽ được hiệu chỉnh sao cho mỗi thành phần của nó được cách nhau đúng một dấu cách, và thêm một dấu cách vào cuối cho dễ tách các phần tử ra để xử lý. Vì Stack chỉ dùng để chứa các toán tử và dấu ngoặc mở nên có thể mô tả Stack dưới dạng xâu ký tự cho đơn giản.

Input: File văn bản RPNCONV.INP chỉ gồm 1 dòng chứa biểu thức trung tố.

Output: File văn bản RPNCONV.OUT ghi biểu thức trung tố sau khi đã hiệu chỉnh và biểu thức RPN tương ứng

Ví dụ:

RPNCONV.INP	RPNCONV.OUT
(10/2 + 3)* (7-4)	Refined: (10 / 2 + 3) * (7 - 4) RPN : 10 2 / 3 + 7 4 - *

```
P_2_07_2.PAS * Chuyển biểu thức trung tố sang dạng RPN
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program ConvertInfixToRPN;
const
  InputFile = 'RPNCONV.INP';
  OutputFile = 'RPNCONV.OUT';
  Opt = ['(', ')', '+', '-', '*', '/'];
var
  T, Infix, Stack: string;
  p: Integer;
  f: Text;

procedure StackInit;
begin
  Stack := '';
end;

procedure Push(V: Char); {Đẩy một toán tử vào Stack}
```

```

begin
  Stack := Stack + V;
end;

function Pop: Char; {Lấy một toán tử ra khỏi Stack, trả về trong kết quả hàm}
begin
  Pop := Stack[Length(Stack)];
  Delete(Stack, Length(Stack), 1);
end;

function Get: Char; {Đọc toán tử ở đỉnh Stack}
begin
  Get := Stack[Length(Stack)];
end;

procedure Refine(var S: String); {Hiệu chỉnh biểu thức trung tố}
var
  i: Integer;
begin
  S := S + ' ';
  for i := Length(S) - 1 downto 1 do
    if (S[i] in Opt) or (S[i + 1] in Opt) then
      Insert(' ', S, i + 1);
  for i := Length(S) - 1 downto 1 do
    if (S[i] = ' ') and (S[i + 1] = ' ') then Delete(S, i + 1, 1);
end;

function Priority(Ch: Char): Integer; {Hàm trả về độ ưu tiên của các toán tử và dấu ngoặc mở}
begin
  case ch of
    '*', '/': Priority := 2;
    '+', '-': Priority := 1;
    '(': Priority := 0;
  end;
end;

procedure Process(T: String); {Xử lý một phần tử đọc được từ biểu thức trung tố}
var
  c, x: Char;
begin
  c := T[1];
  case c of
    '(': Push(c); {T là dấu ( thì đẩy T vào Stack)
    ')': repeat {T là dấu ) thì lấy ra và hiển thị các phần tử trong Stack đến khi lấy tới ()}
      x := Pop;
      if x <> '(' then Write(f, x, ' ');
      until x = '(';
    '+', '-', '*', '/': {T là toán tử}
      begin
        while (Stack <> '') and (Priority(c) <= Priority(Get)) do
          Write(f, Pop, ' ');
        Push(c);
      end;
    else
      Write(f, T, ' '); {T là toán hạng thì hiển thị luôn}
    end;
  end;
begin
  Assign(f, InputFile); Reset(f);
  Readln(f, Infix);
  Close(f);
  Assign(f, OutputFile); Rewrite(f);

```

```

Refine(Inf);
Writeln(f, 'Refined: ', Inf);
Write(f, 'RPN      : ');
T := '';
for p := 1 to Length(Inf) do {Tách và xử lý từng phần tử đọc được từ biểu thức trung tố}
  if Inf[p] <> ' ' then T := T + Inf[p]
  else
    begin
      Process(T);
      T := '';
    end;
  while Stack <> '' do Write(f, Pop, ' ');
Close(f);
end.

```

7.5. XÂY DỰNG CÂY NHỊ PHÂN BIỂU DIỄN BIỂU THỨC

Ngay trong phần đầu tiên, chúng ta đã biết rằng các dạng biểu thức trung tố, tiền tố và hậu tố đều có thể được hình thành bằng cách duyệt cây nhị phân biểu diễn biểu thức đó theo các trật tự khác nhau. Vậy tại sao không xây dựng ngay cây nhị phân biểu diễn biểu thức đó rồi thực hiện các công việc tính toán ngay trên cây?. Khó khăn gặp phải chính là thuật toán xây dựng cây nhị phân trực tiếp từ dạng trung tố có thể kém hiệu quả, trong khi đó từ dạng hậu tố lại có thể khôi phục lại cây nhị phân biểu diễn biểu thức một cách rất đơn giản, gần giống như quá trình tính toán biểu thức hậu tố:

Bước 1: Khởi tạo một Stack rỗng dùng để chứa các nút trên cây

Bước 2: Đọc lần lượt các phần tử của biểu thức RPN từ trái qua phải (phần tử này có thể là hằng, biến hay toán tử) với mỗi phần tử đó:

- ❖ Tạo ra một nút mới N chứa phần tử mới đọc được
- ❖ Nếu phần tử này là một toán tử, lấy từ Stack ra hai nút (theo thứ tự là y và x), sau đó đem liên kết trái của N trỏ đến x, đem liên kết phải của N trỏ đến y.
- ❖ Đẩy nút N vào Stack

Bước 3: Sau khi kết thúc bước 2 thì toàn bộ biểu thức đã được đọc xong, trong Stack chỉ còn duy nhất một phần tử, phần tử đó chính là gốc của cây nhị phân biểu diễn biểu thức.

Bài tập

Bài 1

Viết chương trình chuyển biểu thức trung tố dạng phức tạp hơn bao gồm: Phép lấy số đối ($-x$), phép luỹ thừa x^y (x^y), lời gọi hàm số học (sqrt, exp, abs v.v...) sang dạng RPN.

Bài 2

Viết chương trình chuyển biểu thức logic dạng trung tố sang dạng RPN. Ví dụ:

Chuyển: “a and b or c and d” thành: “a b and c d and or”

Bài 3

Chuyển các biểu thức sau đây ra dạng RPN

- a) $A * (B + C)$
- b) $A + B / C + D$

- c) $A * (B + -C)$
- d) $A - (B + C)^{d/e}$
- e) A and B or C
- f) A and (B or not C)
- g) (A or B) and (C or (D and not E))
- h) (A = B) or (C = D)
- i) (A < 9) and (A > 3) or not (A > 0)
- j) ((A > 0) or (A < 0)) and (B * B - 4 * A * C < 0)

Bài 4

Viết chương trình tính biểu thức logic dạng RPN với các toán tử and, or, not và các toán hạng là TRUE hay FALSE.

Bài 5

Viết chương trình hoàn chỉnh tính giá trị biểu thức trung tố.

§8. SẮP XẾP (SORTING)

8.1. BÀI TOÁN SẮP XẾP

Sắp xếp là quá trình bố trí lại các phần tử của một tập đối tượng nào đó theo một thứ tự nhất định. Chẳng hạn như thứ tự tăng dần (hay giảm dần) đối với một dãy số, thứ tự từ điển đối với các từ v.v... Yêu cầu về sắp xếp thường xuyên xuất hiện trong các ứng dụng Tin học với các mục đích khác nhau: sắp xếp dữ liệu trong máy tính để tìm kiếm cho thuận lợi, sắp xếp các kết quả xử lý để in ra trên bảng biểu v.v...

Nói chung, dữ liệu có thể xuất hiện dưới nhiều dạng khác nhau, nhưng ở đây ta quy ước: Một tập các đối tượng cần sắp xếp là tập các bản ghi (records), mỗi bản ghi bao gồm một số trường (fields) khác nhau. Nhưng không phải toàn bộ các trường dữ liệu trong bản ghi đều được xem xét đến trong quá trình sắp xếp mà chỉ là một trường nào đó (hay một vài trường nào đó) được chú ý tới thôi. Trường như vậy ta gọi là **khoá (key)**. Sắp xếp sẽ được tiến hành dựa vào giá trị của khoá này.

Ví dụ: Hồ sơ tuyển sinh của một trường Đại học là một danh sách thí sinh, mỗi thí sinh có tên, số báo danh, điểm thi. Khi muốn liệt kê danh sách những thí sinh trúng tuyển tức là phải sắp xếp các thí sinh theo thứ tự từ điểm cao nhất tới điểm thấp nhất. Ở đây khoá sắp xếp chính là điểm thi.

STT	SBD	Họ và tên	Điểm thi
1	A100	Nguyễn Văn A	20
2	B200	Trần Thị B	25
3	X150	Phạm Văn C	18
4	G180	Đỗ Thị D	21

Khi sắp xếp, các bản ghi trong bảng sẽ được đặt lại vào các vị trí sao cho giá trị khoá tương ứng với chúng có đúng thứ tự đã định. Vì kích thước của toàn bản ghi có thể rất lớn, nên nếu việc sắp xếp thực hiện trực tiếp trên các bản ghi sẽ đòi hỏi sự chuyển đổi vị trí của các bản ghi, kéo theo việc thường xuyên phải di chuyển, copy những vùng nhớ lớn, gây ra những tổn phí thời gian khá nhiều. Thường người ta khắc phục tình trạng này bằng cách xây dựng một bảng khoá: Mỗi bản ghi trong bảng ban đầu sẽ tương ứng với một bản ghi trong **bảng khoá**. Bảng khoá cũng gồm các bản ghi nhưng mỗi bản ghi chỉ gồm có hai trường:

Trường thứ nhất chứa khoá

Trường thứ hai liên kết tới một bản ghi trong bảng ban đầu, tức là chứa một thông tin đủ để biết bản ghi tương ứng với nó trong bảng ban đầu là bản ghi nào.

Sau đó, việc sắp xếp được thực hiện trực tiếp trên bảng khoá, trong quá trình sắp xếp, **bảng chính không hề bị ảnh hưởng gì**, việc truy cập vào một bản ghi nào đó của bảng chính vẫn

có thể thực hiện được bằng cách dựa vào trường liên kết của bản ghi tương ứng thuộc bảng khoá.

Như ở ví dụ trên, ta có thể xây dựng bảng khoá gồm 2 trường, trường khoá chứa điểm và trường liên kết chứa số thứ tự của người có điểm tương ứng trong bảng ban đầu:

Điểm thi	STT
20	1
25	2
18	3
21	4

Sau khi sắp xếp theo trật tự điểm cao nhất tới điểm thấp nhất, bảng khoá sẽ trở thành:

Điểm thi	STT
25	2
21	4
20	1
18	3

Dựa vào bảng khoá, ta có thể biết được rằng người có điểm cao nhất là người mang số thứ tự 2, tiếp theo là người mang số thứ tự 4, tiếp nữa là người mang số thứ tự 1, và cuối cùng là người mang số thứ tự 3, còn muốn liệt kê danh sách đầy đủ thì ta chỉ việc đổi chiều với bảng ban đầu và liệt kê theo thứ tự 2, 4, 1, 3.

Có thể còn cải tiến tốt hơn dựa vào nhận xét sau: Trong bảng khoá, nội dung của trường khoá hoàn toàn có thể suy ra được từ trường liên kết bằng cách: Dựa vào trường liên kết, tìm tới bản ghi tương ứng trong bảng chính rồi truy xuất trường khoá trong bảng chính. Như ví dụ trên thì người mang số thứ tự 1 chắc chắn sẽ phải có điểm thi là 20, còn người mang số thứ tự 3 thì chắc chắn phải có điểm thi là 18. Vậy thì bảng khoá có thể loại bỏ đi trường khoá mà chỉ giữ lại trường liên kết. Trong trường hợp các phần tử trong bảng ban đầu được đánh số từ 1 tới n và trường liên kết chính là số thứ tự của bản ghi trong bảng ban đầu như ở ví dụ trên, người ta gọi kỹ thuật này là kỹ thuật **sắp xếp bằng chỉ số**: Bảng ban đầu không hề bị ảnh hưởng gì cả, việc sắp xếp chỉ đơn thuần là đánh lại chỉ số cho các bản ghi theo thứ tự sắp xếp.

Cụ thể hơn:

Nếu $r[1..n]$ là các bản ghi cần sắp xếp theo một thứ tự nhất định thì việc sắp xếp bằng chỉ số tức là xây dựng một dãy $Index[1..n]$ mà ở đây:

$$Index[j] = \text{Chỉ số của bản ghi sẽ đứng thứ } j \text{ khi sắp xếp}$$

(Bản ghi $r[Index[j]]$ sẽ phải đứng sau $j - 1$ bản ghi khác khi sắp xếp)

Do khoá có vai trò đặc biệt như vậy nên sau này, khi trình bày các giải thuật, ta sẽ coi **khoá như đại diện cho các bản ghi** và để cho đơn giản, ta chỉ nói tới giá trị của khoá mà thôi. Các thao tác trong kỹ thuật sắp xếp lẽ ra là tác động lên toàn bản ghi giờ đây chỉ làm trên khoá. Còn việc cài đặt các phương pháp sắp xếp trên danh sách các bản ghi và kỹ thuật sắp xếp bằng chỉ số, ta coi như bài tập.

Bài toán sắp xếp giờ đây có thể phát biểu như sau:

Xét quan hệ thứ tự toàn phần “nhỏ hơn hoặc bằng” ký hiệu “ \leq ” trên một tập hợp S, là quan hệ hai ngôi thoả mãn bốn tính chất:

Với $\forall a, b, c \in S$

- ❖ Tính phô biến: Hoặc là $a \leq b$, hoặc $b \leq a$;
- ❖ Tính phản xạ: $a \leq a$
- ❖ Tính phản đối xứng: Nếu $a \leq b$ và $b \leq a$ thì bắt buộc $a = b$.
- ❖ Tính bắc cầu: Nếu có $a \leq b$ và $b \leq c$ thì $a \leq c$.

Trong trường hợp $a \leq b$ và $a \neq b$, ta dùng ký hiệu “ $<$ ” cho gọn

Cho một dãy $k[1..n]$ gồm n khoá. Giữa hai khoá bất kỳ có quan hệ thứ tự toàn phần “ \leq ”. Xếp lại dãy các khoá đó để được dãy khoá thoả mãn $k[1] \leq k[2] \leq \dots \leq k[n]$.

Giả sử cấu trúc dữ liệu cho dãy khoá được mô tả như sau:

```
const
  n = ...; {Số khoá trong dãy khoá, có thể khai báo dưới dạng biến số nguyên để tùy biến hơn}
type
  TKey = ...; {Kiểu dữ liệu một khoá}
  TArray = array[1..n] of TKey;
var
  k: TArray; {Dãy khoá}
```

Thì những thuật toán sắp xếp dưới đây được viết dưới dạng thủ tục sắp xếp dãy khoá k, kiểu chỉ số đánh cho từng khoá trong dãy có thể coi là số nguyên Integer.

8.2. THUẬT TOÁN SẮP XẾP KIỂU CHỌN (SELECTIONSORT)

Một trong những thuật toán sắp xếp đơn giản nhất là phương pháp sắp xếp kiểu chọn. Ý tưởng cơ bản của cách sắp xếp này là:

Ở lượt thứ nhất, ta chọn trong dãy khoá $k[1..n]$ ra khoá nhỏ nhất (khoá \leq mọi khoá khác) và đổi giá trị của nó với $k[1]$, khi đó giá trị khoá $k[1]$ trở thành giá trị khoá nhỏ nhất.

Ở lượt thứ hai, ta chọn trong dãy khoá $k[2..n]$ ra khoá nhỏ nhất và đổi giá trị của nó với $k[2]$.

...

Ở lượt thứ i, ta chọn trong dãy khoá $k[i..n]$ ra khoá nhỏ nhất và đổi giá trị của nó với $k[i]$.

...

Làm tới lượt thứ $n - 1$, chọn trong hai khoá $k[n-1], k[n]$ ra khoá nhỏ nhất và đổi giá trị của nó với $k[n-1]$.

```

procedure SelectionSort;
var
  i, j, jmin: Integer;
begin
  for i := 1 to n - 1 do {Làm n - 1 lượt}
    begin
      {Chọn trong số các khoá trong đoạn k[i..n] ra khoá k[jmin] nhỏ nhất}
      jmin := i;
      for j := i + 1 to n do
        if k[j] < k[jmin] then jmin := j;
      if jmin ≠ i then
        <Đảo giá trị của k[jmin] cho k[i]>
    end;
end;

```

Đối với phương pháp kiểu lựa chọn, có thể coi phép so sánh ($k[j] < k[jmin]$) là phép toán tích cực để đánh giá hiệu suất thuật toán về mặt thời gian. Ở lượt thứ i , để chọn ra khoá nhỏ nhất bao giờ cũng cần $n - i$ phép so sánh, số lượng phép so sánh này không hề phụ thuộc gì vào tình trạng ban đầu của dãy khoá cả. Từ đó suy ra tổng số phép so sánh sẽ phải thực hiện là:

$$(n - 1) + (n - 2) + \dots + 1 = n * (n - 1) / 2$$

Vậy thuật toán sắp xếp kiểu chọn có độ phức tạp tính toán là $O(n^2)$

8.3. THUẬT TOÁN SẮP XẾP NỐI BỌT (BUBBLE SORT)

Trong thuật toán sắp xếp nối bọt, dãy các khoá sẽ được duyệt từ cuối dãy lên đầu dãy (từ $k[n]$ về $k[1]$), nếu gặp hai khoá kế cận bị ngược thứ tự thì đổi chỗ của chúng cho nhau. Sau lần duyệt như vậy, khoá nhỏ nhất trong dãy khoá sẽ được chuyển về vị trí đầu tiên và vấn đề trở thành sắp xếp dãy khoá từ $k[2]$ tới $k[n]$:

```

procedure BubbleSort;
var
  i, j: Integer;
begin
  for i := 2 to n do
    for j := n downto i do {Duyệt từ cuối dãy lên, làm nổi khoá nhỏ nhất trong đoạn k[i-1, n] về vị trí i-1}
      if k[j] < k[j-1] then
        <Đảo giá trị k[j] và k[j-1]>
end;

```

Đối với thuật toán sắp xếp nối bọt, có thể coi phép toán tích cực là phép so sánh $k[j] < k[j-1]$.

Và số lần thực hiện phép so sánh này là:

$$(n - 1) + (n - 2) + \dots + 1 = n * (n - 1) / 2$$

Vậy thuật toán sắp xếp nối bọt cũng có độ phức tạp là $O(n^2)$. Bất kể tình trạng dữ liệu vào như thế nào.

8.4. THUẬT TOÁN SẮP XẾP KIỂU CHÈN (INSERTION SORT)

Xét dãy khoá $k[1..n]$. Ta thấy dãy con chỉ gồm mỗi một khoá là $k[1]$ có thể coi là đã sắp xếp rồi. Xét thêm $k[2]$, ta so sánh nó với $k[1]$, nếu thấy $k[2] < k[1]$ thì chèn nó vào trước $k[1]$. Đối với $k[3]$, ta lại xét dãy chỉ gồm 2 khoá $k[1], k[2]$ đã sắp xếp và tìm cách chèn $k[3]$ vào dãy khoá đó để được thứ tự sắp xếp. Một cách tổng quát, ta sẽ sắp xếp dãy $k[1..i]$ trong điều kiện dãy $k[1..i-1]$ đã sắp xếp rồi bằng cách chèn $k[i]$ vào dãy đó tại vị trí đúng khi sắp xếp.

```

procedure InsertionSort;
var
  i, j: Integer;
  tmp: TKey; {Biến giữ lại giá trị khoá chèn}
begin
  for i := 2 to n do {Chèn giá trị k[i] vào dãy k[1..i-1] để toàn đoạn k[1..i] trở thành đã sắp xếp}
    begin
      tmp := k[i]; {Giữ lại giá trị k[i]}
      j := i - 1;
      while (j > 0) and (tmp < k[j]) do {So sánh giá trị cần chèn với lần lượt các khoá k[j] (i-1≥j≥0)}
        begin
          k[j+1] := k[j]; {Đẩy lùi giá trị k[j] về phía sau một vị trí, tạo ra "khoảng trống" tại vị trí j}
          j := j - 1;
        end;
      k[j+1] := tmp; {Đưa giá trị chèn vào "khoảng trống" mới tạo ra}
    end;
  end;
end;

```

Đối với thuật toán sắp xếp kiểu chèn, thì chi phí thời gian thực hiện thuật toán phụ thuộc vào tình trạng dãy khoá ban đầu. Nếu coi phép toán tích cực ở đây là phép so sánh $\text{tmp} < k[j]$, ta có:

Trường hợp tốt nhất ứng với dãy khoá đã sắp xếp rồi, mỗi lượt chỉ cần 1 phép so sánh, và như vậy tổng số phép so sánh được thực hiện là $n - 1$. Phân tích trong trường hợp tốt nhất, độ phức tạp tính toán của InsertionSort là $\Theta(n)$

Trường hợp tồi tệ nhất ứng với dãy khoá đã có thứ tự ngược với thứ tự cần sắp thì ở lượt thứ i, cần có $i - 1$ phép so sánh và tổng số phép so sánh là:

$$(n - 1) + (n - 2) + \dots + 1 = n * (n - 1) / 2.$$

Vậy phân tích trong trường hợp tốt nhất, độ phức tạp tính toán của InsertionSort là $\Theta(n^2)$

Trường hợp các giá trị khoá xuất hiện một cách ngẫu nhiên, ta có thể coi xác suất xuất hiện mỗi khoá là đồng khả năng, thì có thể coi ở lượt thứ i, thuật toán cần trung bình $i / 2$ phép so sánh và tổng số phép so sánh là:

$$(1 / 2) + (2 / 2) + \dots + (n / 2) = (n + 1) * n / 4.$$

Vậy phân tích trong trường hợp trung bình, độ phức tạp tính toán của InsertionSort là $\Theta(n^2)$.

Nhìn về kết quả đánh giá, ta có thể thấy rằng thuật toán sắp xếp kiểu chèn tỏ ra tốt hơn so với thuật toán sắp xếp chọn và sắp xếp nổi bọt. Tuy nhiên, chi phí thời gian thực hiện của thuật toán sắp xếp kiểu chèn vẫn còn khá lớn.

Có thể cải tiến thuật toán sắp xếp chèn nhờ nhận xét: Khi dãy khoá $k[1..i-1]$ đã được sắp xếp thì việc tìm vị trí chèn có thể làm bằng thuật toán tìm kiếm nhị phân và kỹ thuật chèn có thể làm bằng các lệnh dịch chuyển vùng nhớ cho nhanh. Tuy nhiên điều đó cũng không làm giảm đi độ phức tạp của thuật toán bởi trong trường hợp xấu nhất, ta phải mất $n - 1$ lần chèn và lần chèn thứ i ta phải dịch lùi i khoá để tạo ra khoảng trống trước khi đẩy giá trị khoá chèn vào chỗ trống đó.

```

procedure InsertionSortwithBinarySearching;
var
  i, inf, sup, median: Integer;
  tmp: TKey;
begin
  for i := 2 to n do
    begin
      tmp := k[i]; {Giữ lại giá trị k[i]}
      inf := 1; sup := i - 1; {Tim chỗ chèn giá trị tmp vào đoạn từ k[inf] tới k[sup+1]}
      repeat {Sau mỗi vòng lặp này thì đoạn tìm bị co lại một nửa}
        median := (inf + sup) div 2; {Xét chỉ số nằm giữa chỉ số inf và chỉ số sup}
        if tmp < k[median] then sup := median - 1
        else inf := median + 1;
      until inf > sup; {Kết thúc vòng lặp thì inf = sup + 1 chính là vị trí chèn}
      <Địch các khoá từ k[inf] tới k[i-1] lùi sau một vị trí>
      k[inf] := tmp; {Đưa giá trị tmp vào "khoảng trống" mới tạo ra}
    end;
end;

```

8.5. SẮP XẾP CHÈN VỚI ĐỘ DÀI BƯỚC GIẢM DÀN (SHELLSORT)

Nhược điểm của thuật toán sắp xếp kiểu chèn thể hiện khi mà ta luôn phải chèn một khóa vào vị trí gần đầu dãy. Để khắc phục nhược điểm này, người ta thường sử dụng thuật toán sắp xếp chèn với độ dài bước giảm dần, ý tưởng ban đầu cho thuật toán được đưa ra bởi D.L.Shell năm 1959 nên thuật toán còn có một tên gọi khác: ShellSort

Xét dãy khoá: $k[1..n]$. Với một số nguyên dương h : $1 \leq h \leq n$, ta có thể chia dãy đó thành h dãy con:

Dãy con 1: $k[1], k[1+h], k[1 + 2h], \dots$

Dãy con 2: $k[2], k[2+h], k[2 + 2h], \dots$

...

Dãy con h : $k[h], k[2h], k[3h], \dots$

Ví dụ như dãy $(4, 6, 7, 2, 3, 5, 1, 9, 8)$; $n = 9$; $h = 3$. Có 3 dãy con.

Dãy khoá chính:	4	6	7	2	3	5	1	9	8
Dãy con 1:	4			2			1		
Dãy con 2:		6			3			9	
Dãy con 3:			7			5			8

Những dãy con như vậy được gọi là dãy con xếp theo độ dài bước h . Tư tưởng của thuật toán ShellSort là: Với một bước h , áp dụng thuật toán sắp xếp kiểu chèn từng dãy con độc lập để làm mịn dần dãy khoá chính. Rồi lại làm tương tự đối với bước $h \text{ div } 2 \dots$ cho tới khi $h = 1$ thì ta được dãy khoá sắp xếp.

Như ở ví dụ trên, nếu dùng thuật toán sắp xếp kiểu chèn thì khi gặp khoá $k[7] = 1$, là khoá nhỏ nhất trong dãy khoá, nó phải chèn vào vị trí 1, tức là phải thao tác trên 6 khoá đứng trước nó. Nhưng nếu coi 1 là khoá của dãy con 1 thì nó chỉ cần chèn vào trước 2 khoá trong dãy con đó

mà thôi. Đây chính là nguyên nhân ShellSort hiệu quả hơn sáp xếp chèn: Khoá nhỏ được nhanh chóng đưa về **gần** vị trí đúng của nó.

```
procedure ShellSort;
var
  i, j, h: Integer;
  tmp: TKey;
begin
  h := n div 2;
  while h <> 0 do {Làm mịn dãy với độ dài bước h}
    begin
      for i := h + 1 to n do
        begin {Sắp xếp chèn trên dãy con a[i-h], a[i], a[i+h], a[i+2h], ...}
          tmp := k[i]; j := i - h;
          while (j > 0) and (k[j] > tmp) do
            begin
              k[j+h] := k[j];
              j := j - h;
            end;
          k[j+h] := tmp;
        end;
      h := h div 2;
    end;
end;
```

Trên đây là phiên bản nguyên thuỷ của ShellSort do D.L.Shell đưa ra năm 1959. Độ dài bước được đếm div 2 sau mỗi lần lặp. Để thấy rằng để ShellSort hoạt động đúng thì chỉ cần dãy bước h giảm dần về 1 sau mỗi bước lặp là được, đã có một số nghiên cứu về việc chọn dãy bước h cho ShellSort nhằm tăng hiệu quả của thuật toán.

ShellSort hoạt động nhanh và dễ cài đặt, tuy vậy việc đánh giá độ phức tạp tính toán của ShellSort là tương đối khó, ta chỉ thừa nhận các kết quả sau đây:

Nếu các bước h được chọn theo thứ tự ngược từ dãy: 1, 3, 7, 15, ..., $2^i - 1$, ... thì độ phức tạp tính toán của ShellSort là $O(n^{3/2})$.

Nếu các bước h được chọn theo thứ tự ngược từ dãy: 1, 8, 23, 77, ..., $4^{i+1} + 3 \cdot 2^i + 1$, ... thì độ phức tạp tính toán của ShellSort là $O(n^{4/3})$.

Nếu các bước h được chọn theo thứ tự ngược từ dãy: 1, 2, 3, 4, 6, 8, 9, 12, 16, ..., $2^i 3^j$, ... (Dãy tăng dần của các phần tử dạng $2^i 3^j$) thì độ phức tạp tính toán của ShellSort là $O(n(\log n)^2)$.

8.6. THUẬT TOÁN SẮP XẾP KIỀU PHÂN ĐOẠN (QUICKSORT)

8.6.1. Tư tưởng của QuickSort

QuickSort - thuật toán được đề xuất bởi C.A.R. Hoare - là một phương pháp sáp xếp tốt nhất, nghĩa là dù dãy khoá thuộc kiểu dữ liệu có thứ tự nào, QuickSort cũng có thể sáp xếp được và chưa có một thuật toán sáp xếp tổng quát nào nhanh hơn QuickSort về mặt tốc độ trung bình (theo tôi biết). Hoare đã mạnh dạn lấy chữ “Quick” để đặt tên cho thuật toán.

Ý tưởng chủ đạo của phương pháp có thể tóm tắt như sau: Sắp xếp dãy khoá $k[1..n]$ thì có thể coi là sáp xếp đoạn từ chỉ số 1 tới chỉ số n trong dãy khoá đó. Để sáp xếp một đoạn trong dãy khoá, nếu đoạn đó có ít hơn 2 khoá thì không cần phải làm gì cả, còn nếu đoạn đó có ít nhất 2

khoá, ta chọn một khoá ngẫu nhiên nào đó của đoạn làm “chốt” (Pivot). Mọi khoá nhỏ hơn khoá chốt được xếp vào vị trí đứng trước chốt, mọi khoá lớn hơn khoá chốt được xếp vào vị trí đứng sau chốt. Sau phép hoán chuyển như vậy thì đoạn đang xét được chia làm hai đoạn khác rỗng mà mọi khoá trong đoạn đầu đều \leq chốt và mọi khoá trong đoạn sau đều \geq chốt. Hay nói cách khác: Mọi khoá trong đoạn đầu đều \leq mọi khoá trong đoạn sau. Và vấn đề trở thành sắp xếp hai đoạn mới tạo ra (có độ dài ngắn hơn đoạn ban đầu) bằng phương pháp tương tự.


```

procedure QuickSort;

procedure Partition(L, H: Integer); {Sắp xếp dãy khoá k[L..H]}
var
  i, j: Integer;
  Pivot: TKey; {Biến lưu giá trị khoá chốt}
begin
  if L > H then Exit; {Nếu đoạn chỉ có ≤ 1 khoá thì không phải làm gì cả}
  Pivot := k[Random(H - L + 1) + L]; {Chọn một khoá ngẫu nhiên trong đoạn làm khoá chốt}
  i := L; j := H; {i := vị trí đầu đoạn; j := vị trí cuối đoạn}
  repeat
    while k[i] < Pivot do i := i + 1; {Tìm từ đầu đoạn khoá ≥ khoá chốt}
    while k[j] > Pivot do j := j - 1; {Tìm từ cuối đoạn khoá ≤ khoá chốt}
    {Đến đây ta tìm được hai khoá k[i] và k[j] mà k[i] ≥ key ≥ k[j]}
    if i ≤ j then
      begin
        if i < j then {Nếu chỉ số i đúng trước chỉ số j thì đảo giá trị hai khoá k[i] và k[j]}
          {Đảo giá trị k[i] và k[j]}; {Sau phép đảo này ta có: k[i] ≤ key ≤ k[j]}
        i := i + 1; j := j - 1;
      end;
    until i > j;
    Partition(L, i); Partition(i, H); {Sắp xếp hai đoạn con mới tạo ra}
  end;

begin
  Partition(1, n);
end;
```

Ta thử phân tích xem tại sao đoạn chương trình trên hoạt động đúng: Xét vòng lặp repeat...until trong lần lặp đầu tiên, vòng lặp while thứ nhất chắc chắn sẽ tìm được khoá $k[i] \geq$ khoá chốt bởi chắc chắn tồn tại trong đoạn một khoá bằng khoá chốt. Tương tự như vậy, vòng lặp while thứ hai chắc chắn tìm được khoá $k[j] \leq$ khoá chốt. Nếu như khoá $k[i]$ đứng trước khoá $k[j]$ thì ta đảo giá trị hai khoá, cho i tiến và j lùi. Khi đó ta có nhận xét rằng mọi khoá đứng trước vị trí i sẽ phải \leq khoá chốt và mọi khoá đứng sau vị trí j sẽ phải \geq khoá chốt.




Hình 29: Vòng lặp trong của QuickSort

Điều này đảm bảo cho vòng lặp repeat...until tại bước sau, hai vòng lặp while...do bên trong chắc chắn lại tìm được hai khoá $k[i]$ và $k[j]$ mà $k[i] \geq$ khoá chốt $\geq k[j]$, nếu khoá $k[i]$ đứng

trước khoá $k[j]$ thì lại đảo giá trị của chúng, cho i tiến lên một vị trí và j lùi về một vị trí. Vậy vòng lặp repeat...until sẽ đảm bảo tại mỗi bước:

- ❖ Hai vòng lặp while...do bên trong luôn tìm được hai khoá $k[i]$, $k[j]$ mà $k[i] \geq$ khoá chót $\geq k[j]$. Không có trường hợp hai chỉ số i , j chạy ra ngoài đoạn (luôn luôn có $L \leq i, j \leq H$).
- ❖ Sau mỗi phép hoán chuyển, mọi khoá đứng trước vị trí i luôn \leq khoá chót và mọi khoá đứng sau vị trí j luôn \geq khoá chót.

Vòng lặp repeat ...until sẽ kết thúc khi mà chỉ số i đứng phía sau chỉ số j (Hình 30).



Hình 30: Trạng thái trước khi gọi đệ quy

Theo những nhận xét trên, nếu có một khoá nằm giữa $k[j]$ và $k[i]$ thì khoá đó phải đúng bằng khoá chót và nó đã được đặt ở vị trí đúng của nó, nên có thể bỏ qua khoá này mà chỉ xét hai đoạn ở hai đầu. Công việc còn lại là gọi đệ quy để làm tiếp với đoạn từ $k[L]$ tới $k[j]$ và đoạn từ $k[i]$ tới $k[H]$. Hai đoạn này ngắn hơn đoạn đang xét bởi vì $L \leq j < i \leq H$. Vậy thuật toán không bao giờ bị rơi vào quá trình vô hạn mà sẽ dừng và cho kết quả đúng đắn.

Xét về độ phức tạp tính toán, trường hợp tốt nhất là tại mỗi bước chọn chốt để phân đoạn, ta chọn đúng trung vị của dãy khoá (giá trị sẽ đứng giữa dãy khi sắp thứ tự), khi đó độ phức tạp tính toán của QuickSort là $\Theta(n \lg n)$. Trường hợp tồi tệ nhất là tại mỗi bước chọn chốt để phân đoạn, ta chọn đúng vào khoá lớn nhất hoặc nhỏ nhất của dãy khoá, tạo ra một đoạn gồm 1 khoá và đoạn còn lại gồm $n - 1$ khoá, khi đó độ phức tạp tính toán của QuickSort là $\Theta(n^2)$. Thời gian thực hiện giải thuật QuickSort trung bình là $\Theta(n \lg n)$. Việc chứng minh các kết quả này phải sử dụng những công cụ toán học phức tạp, ta thưa nhận những điều nói trên.

8.6.2. Trung vị và thứ tự thống kê (median and order statistics)

Việc chọn chốt cho phép phân đoạn quyết định hiệu quả của QuickSort, nếu chọn chốt không tốt, rất có thể việc phân đoạn bị suy biến thành trường hợp xấu khiến QuickSort hoạt động chậm và tràn ngắn xếp chương trình con khi gặp phải dây chuyền đệ qui quá dài. Những ví dụ sau đây cho thấy với một chiến lược chọn chốt tồi có thể dễ dàng tìm ra những bộ dữ liệu khiến QuickSort hoạt động chậm.

Với m khá lớn:

- ❖ Nếu như chọn chốt là khoá đầu đoạn ($Pivot := k[L]$) hay chọn chốt là khoá cuối đoạn ($Pivot := k[H]$) thì QuickSort sẽ trở thành “Slow” Sort với dãy $(1, 2, \dots, m)$.
- ❖ Nếu như chọn chốt là khoá giữa đoạn ($Pivot := k[(L+H) \text{ div } 2]$) thì QuickSort cũng trở thành “Slow” Sort với dãy $(1, 2, \dots, m-1, m, m-1, \dots, 2, 1)$.

- ❖ Trong trường hợp chọn chốt là khoá nằm ở vị trí ngẫu nhiên trong đoạn, thật khó có thể tìm ra một bộ dữ liệu khiến cho QuickSort hoạt động chậm. Nhưng ta cũng cần hiểu rằng với mọi thuật toán tạo số ngẫu nhiên, trong $m!$ dãy hoán vị của dãy $(1, 2, \dots, m)$ thế nào cũng có một dãy làm QuickSort bị suy biến, tuy nhiên xác suất xảy ra dãy này quá nhỏ và cũng rất khó để chỉ ra nên việc sử dụng cách chọn chốt là khoá nằm ở vị trí ngẫu nhiên có thể coi là an toàn với các trường hợp suy biến của QuickSort.

Phần “trung vị và thứ tự thống kê” này được trình bày trong nội dung thảo luận về QuickSort bởi nó cung cấp một chiến lược chọn chốt “đẹp” trên lý thuyết, nghĩa là trong trường hợp xấu nhất, độ phức tạp tính toán của QuickSort cũng chỉ là $O(n \lg n)$ mà thôi. Để giải quyết vấn đề suy biến của QuickSort, ta xét bài toán tìm trung vị của dãy khoá và bài toán tổng quát hơn: Bài toán thứ tự thống kê (Order statistics).

Bài toán: Cho dãy khoá k_1, k_2, \dots, k_n , hãy chỉ ra khoá sẽ đứng thứ p trong dãy khi sắp thứ tự. Khi $p = n \text{ div } 2$ thì bài toán thứ tự thống kê trở thành bài toán tìm trung vị của dãy khoá. Sau đây ta sẽ nói về một số cách giải quyết bài toán thứ tự thống kê với mục tiêu cuối cùng là tìm ra một thuật toán để giải bài toán này với độ phức tạp trong trường hợp xấu nhất là $O(n)$. Cách tệ nhất mà ai cũng có thể nghĩ tới là sắp xếp lại toàn bộ dãy k và đưa ra khoá đứng thứ p của dãy đã sắp. Trong các thuật toán sắp xếp tổng quát mà ta thảo luận trong bài, không thuật toán nào cho phép thực hiện việc này với độ phức tạp xấu nhất và trung bình là $O(n)$ cả.

Cách thứ hai là sửa đổi một chút thủ tục Partition của QuickSort: thủ tục Partition chọn khoá chốt và chia đoạn đang xét làm hai đoạn con (thực ra là ba): Các khoá của đoạn đầu \leq chốt, các khoá của đoạn giữa = chốt, các khoá của đoạn sau \geq chốt. Khi đó ta hoàn toàn có thể xác định được khoá cần tìm nằm ở đoạn nào. Nếu khoá đó nằm ở đoạn giữa thì ta chỉ việc trả về giá trị khoá chốt. Nếu khoá đó nằm ở đoạn đầu hay đoạn sau thì chỉ cần gọi đệ quy làm tương tự với một trong hai đoạn đó chứ không cần gọi đệ quy để sắp xếp cả hai đoạn như QuickSort.

```

{
Input: Dãy khoá k[1..n], số p (1 ≤ p ≤ n)
Output: Giá trị khoá đứng thứ p trong dãy sau khi sắp thứ tự được trả về trong lời gọi hàm Select(l, n)
}
function Select(L, H: Integer): TKey; {Tim trong đoạn k[L..H]}
var
  Pivot: TKey;
  i, j: Integer;
begin
  Pivot := k[Random(H - L + 1) + L];
  i := L; j := H;
  repeat
    while k[i] < Pivot do i := i + 1;
    while k[j] > Pivot do j := j - 1;
    if i ≤ j then
      begin
        if i < j then {Đảo giá trị k[i] và k[j]};
        i := i + 1; j := j - 1;
      end;
    until i > j;
  {Xác định khoá cần tìm nằm ở đoạn nào}
  if p ≤ j then Select := Select(L, j) {Khoá cần tìm nằm trong đoạn đầu}
  else
    if p ≥ i then Select := Select(i, H) {Khoá cần tìm nằm trong đoạn sau}
    else Select := Pivot; {Khoá cần tìm nằm ở đoạn giữa, chỉ cần trả về Pivot}
end;

```

Cách thứ hai tốt hơn cách thứ nhất khi phân tích độ phức tạp trung bình về thời gian thực hiện giải thuật (Có thể chứng minh được là $O(n)$). Tuy nhiên trong trường hợp xấu nhất, giải thuật này vẫn có độ phức tạp $O(n^2)$ khi cần chỉ ra khoá lớn nhất của dãy khoá và chốt Pivot được chọn luôn là khoá nhỏ nhất của đoạn $k[L..H]$. Ta vẫn phải hướng tới một thuật toán tốt hơn nữa.

Cách thứ ba: Sự bí hiểm của số 5.

Ta sẽ viết một hàm $Select(L, H, p)$ trả về khoá sẽ đứng thứ p khi sắp xếp dãy khoá $k[L..H]$. Nếu dãy này có ít hơn 50 khoá, thuật toán sắp xếp kiểu chèn sẽ được áp dụng trên dãy khoá này và sau đó giá trị $k[L + p - 1]$ sẽ được trả về trong kết quả hàm $Select$.

Nếu dãy này có ≥ 50 khoá, ta chia các khoá $k[L..H]$ thành các nhóm 5 khoá:

$k[L + 0..L + 4], k[L + 5..L + 9], k[L + 10, L + 14]...$

Nếu cuối cùng quá trình chia nhóm còn lại ít hơn 5 khoá (do độ dài đoạn $k[L..H]$ không chia hết cho 5), ta bỏ qua không xét những khoá dư thừa này.

Với mỗi nhóm 5 khoá kể trên, ta tìm trung vị của nhóm (gọi tắt là trung vị nhóm - khoá đứng thứ 3 khi sắp thứ tự 5 khoá) và đẩy trung vị nhóm ra đầu đoạn $k[L..H]$ theo thứ tự:

Trung vị của $k[L + 0..L + 4]$ sẽ được đảo giá trị cho $k[L]$

Trung vị của $k[L + 5..L + 9]$ sẽ được đảo giá trị cho $k[L + 1]$

...

Giả sử trung vị của nhóm cuối cùng sẽ được đảo giá trị cho $k[j]$.

Sau khi các trung vị nhóm đã tập trung về các vị trí $k[L..j]$, ta đặt Pivot bằng trung vị của các trung vị nhóm bằng một lệnh gọi đệ quy hàm $Select$:

Pivot := Select(L, j, (j - L + 1) div 2);

Tiếp tục các lệnh của hàm Select như thế nào sẽ bàn sau, bây giờ ta giả sử hàm Select hoạt động đúng để xét một tính chất quan trọng của Pivot:

Nếu độ dài đoạn $k[L..H]$ là η ($= H - L + 1$) thì có η div 5 nhóm, nên cũng có η div 5 trung vị nhóm. Pivot là trung vị của các trung vị nhóm nên Pivot phải lớn hơn hay bằng (η div 5) div 2 trung vị nhóm, mỗi trung vị nhóm lại lớn hơn hay bằng 2 khoá khác của nhóm. Vậy có thể suy ra rằng Pivot lớn hơn hay bằng (η div 5 div 2 * 3) khoá của đoạn $k[L..H]$. Lập luận tương tự, ta có Pivot nhỏ hơn hay bằng (η div 5 div 2 * 3) khoá khác của đoạn $k[L..H]$. Với $n \geq 50$, ta có η div 5 div 2 * 3 $\geq \eta/4$. Suy ra:

- ❖ Có ít nhất $\eta/4$ khoá nhỏ hơn hay bằng Pivot \Rightarrow có nhiều nhất $3\eta/4$ khoá lớn hơn Pivot
- ❖ Có ít nhất $\eta/4$ khoá lớn hơn hay bằng Pivot \Rightarrow có nhiều nhất $3\eta/4$ khoá nhỏ hơn Pivot

Ta quay lại xây dựng tiếp hàm Select, khi đã có Pivot, ta có thể đếm được bao nhiêu khoá trong đoạn $k[L..H]$ nhỏ hơn Pivot, bao nhiêu khoá bằng Pivot và bao nhiêu khoá lớn hơn Pivot, từ đó xác định được giá trị cần tìm nhỏ hơn, lớn hơn, hay bằng Pivot. Nếu giá trị cần tìm bằng Pivot thì chỉ cần trả về Pivot trong kết quả hàm. Nếu giá trị cần tìm nhỏ hơn Pivot, ta dồn tất cả các khoá nhỏ hơn Pivot trong đoạn $k[L..H]$ về đầu đoạn và gọi đệ quy tìm tiếp với đoạn đầu này (Chú ý rằng độ dài đoạn được xét tiếp trong lời gọi đệ quy không quá $3/4$ lần độ dài đoạn $k[L..H]$), vẫn đề tương tự đối với trường hợp giá trị cần tìm lớn hơn Pivot.

```

procedure InsertionSort(L, H: Integer);
begin
  {Dùng InsertionSort sắp xếp dãy k[L..H];}
end;

function Select(L, H, p: Integer): TKey; {Hàm trả về khoá nhỏ thứ p trong dãy khoá k[L..H]}
var
  i, j, cL, cE: Integer;
  Pivot: TKey;
begin
  if H - L < 49 then {Nếu độ dài đoạn ít hơn 50 khoá}
    begin
      InsertionSort(L, H); {Thực hiện sắp xếp chèn}
      Select := k[L + p - 1]; {Và trả về phần tử nhỏ thứ p}
      Exit;
    end;
  j := L - 1; i := L;
  repeat {Tim trung vị của k[i, i + 4] chuyển về đầu đoạn}
    InsertionSort(i, i + 4);
    j := j + 1;
    {Đảo giá trị k[i + 2] cho k[j];}
    i := i + 5;
  until i + 5 > H;
  Pivot := Select(L, j, (j - L + 1) div 2);
  cL := 0; cE := 0; {đếm cL: số phần tử nhỏ hơn Pivot, cE: số phần tử bằng Pivot trong dãy k[L, H]}
  for i := L to H do
    if k[i] < Pivot then cL := cL + 1;
    else if k[i] = Pivot then cE := cE + 1;
  if (cL < p) and (p <= cL + cE) then {Giá trị cần tìm bằng Pivot}
    begin
      Select := Pivot;
      Exit;
    end;
  j := L - 1;
  if p <= cL then {Giá trị cần tìm nhỏ hơn Pivot}
    begin
      for i := L to H do {Đồn các khoá nhỏ hơn Pivot về đầu đoạn}
        if k[i] < Pivot then
          begin
            j := j + 1;
            {Đảo giá trị k[i] cho k[j];}
          end;
      Select := Select(L, j, p); {Gọi đệ quy tìm tiếp trong đoạn k[L..j]}
    end
  else {Giá trị cần tìm lớn hơn Pivot}
    begin
      for i := L to H do {Đồn các khoá lớn hơn Pivot về đầu đoạn}
        if k[i] > Pivot then
          begin
            j := j + 1;
            {Đảo giá trị k[i] cho k[j];}
          end;
      Select := Select(L, j, p - cL - cE); {Gọi đệ quy tìm tiếp trong đoạn k[L..j]}
    end;
  end;
end;

```

Ta sẽ chỉ ra rằng độ phức tạp tính toán của thuật toán trên là $O(n)$ trong trường hợp xấu nhất. Nếu gọi $T(n)$ là thời gian thực hiện hàm Select trong trường hợp xấu nhất với độ dài dãy khoá $k[L..H]$ bằng n . Ta có:

$$T(n) \leq \begin{cases} c_1, & \text{if } n \leq 50 \\ c_2 n + T\left(\frac{n}{5}\right) + T\left(\frac{3n}{4}\right), & \text{otherwise} \end{cases}$$

Bởi khi $n \leq 50$ thì thuật toán sắp xếp chèn sẽ được thực hiện, có thể coi đoạn chương trình này kết thúc trong thời gian c_1 với c_1 là một hằng số đủ lớn. Khi $n > 50$, nhìn vào các đoạn mã trong hàm Select, lệnh Pivot := Select(L, j, (j - L + 1) div 2) có thời gian thực hiện $T(n \text{ div } 5)$. Lệnh Select := Select(L, j, ...) có thời gian thực hiện không quá $T(3n/4)$ do tính chất của Pivot. Thời gian thực hiện các lệnh khác trong hàm Select tổng lại có thể coi là không quá $c_2 \cdot n$ với c_2 là một hằng số đủ lớn. Đặt $c = \max(c_1, 20c_2)$, ta có:

Với $1 \leq n \leq 50$, rõ ràng $T(n) \leq c_1 \leq cn$.

Với $n > 50$, giả thiết quy nạp rằng $T(m) \leq cm$ với $\forall m < n$, ta sẽ chứng minh $T(n) \leq cn$, thật vậy:

$$T(n) \leq c_2 n + c \frac{n}{5} + c \frac{3n}{4} \leq c \frac{1}{20} n + c \frac{1}{5} n + c \frac{3}{4} n = cn$$

Ta có điều phải chứng minh: $T(n) = O(n)$. Sự bí ẩn của việc chọn số 5 cho kích thước nhóm đã được giải thích ($1/5 + 3/4 < 1$)

8.6.3. Kết luận:


- ❖ Có thể giải bài toán thứ tự thống kê bằng thuật toán có độ phức tạp $O(n)$ trong trường hợp xấu nhất.
- ❖ Có thể cài đặt thuật toán QuickSort với độ phức tạp $O(nlgn)$ trong trường hợp xấu nhất bởi tại mỗi lần phân đoạn của QuickSort ta có thể tìm được trung vị của dãy trong thời gian $O(n)$ bằng việc giải quyết bài toán thứ tự thống kê
- ❖ Cho tới thời điểm này, khi giải mọi bài toán có chứa thủ tục sắp xếp, ta có thể coi thời gian thực hiện thủ tục sắp xếp đó là $O(nlgn)$ với mọi tình trạng dữ liệu vào.

8.7. THUẬT TOÁN SẮP XẾP KIỂU VŨN ĐỒNG (HEAPSORT)

HeapSort được đề xuất bởi J.W.J. Williams năm 1981, thuật toán không những đóng góp một phương pháp sắp xếp hiệu quả mà còn xây dựng một cấu trúc dữ liệu quan trọng để biểu diễn hàng đợi có độ ưu tiên: Cấu trúc dữ liệu Heap.

8.7.1. Đổng (heap)

Đổng là một dạng cây nhị phân hoàn chỉnh đặc biệt mà giá trị lưu tại mọi nút có độ ưu tiên cao hơn hay bằng giá trị lưu trong hai nút con của nó. Trong thuật toán sắp xếp kiểu vùn đồng, ta coi quan hệ “ưu tiên hơn hay bằng” là quan hệ “lớn hơn hay bằng”: \geq



Hình 31: Heap


8.7.2. Vun đống

Trong bài §6, ta đã biết một dãy khoá $k[1..n]$ là biểu diễn của một cây nhị phân hoàn chỉnh mà $k[i]$ là giá trị lưu trong nút thứ i , nút con của nút thứ i là nút $2i$ và nút $2i + 1$, nút cha của nút thứ j là nút $j \text{ div } 2$. Vấn đề đặt ra là sắp lại dãy khoá đã cho để nó biểu diễn một đống.

Vì cây nhị phân chỉ gồm có một nút hiển nhiên là đống, nên **để vun một nhánh cây gốc r thành đống, ta có thể coi hai nhánh con của nó (nhánh gốc 2r và 2r + 1) đã là đống rồi** và thực hiện thuật toán vun đống từ dưới lên (bottom-up) đối với cây: Gọi h là chiều cao của cây, nút ở mức h (nút lá) đã là gốc một đống, ta vun lên để những nút ở mức $h - 1$ cũng là gốc của đống, ... cứ như vậy cho tới nút ở mức 1 (nút gốc) cũng là gốc của đống.

Thuật toán vun thành đống đối với cây gốc r, hai nhánh con của r đã là đống rồi:

Giả sử ở nút r chứa giá trị V . Từ r , ta cứ đi tới nút con chứa giá trị lớn nhất trong 2 nút con, cho tới khi gặp phải một nút c mà mọi nút con của c đều chứa giá trị $\leq V$ (nút lá cũng là trường hợp riêng của điều kiện này). Dọc trên đường đi từ r tới c , ta đẩy giá trị chứa ở nút con lên nút cha và đặt giá trị V vào nút c .




Hình 32: Vun đống

8.7.3. Tư tưởng của HeapSort


Đầu tiên, dãy khoá $k[1..n]$ được vun từ dưới lên để nó biểu diễn một đống, khi đó khoá $k[1]$ tương ứng với nút gốc của đống là khoá lớn nhất, ta đảo giá trị khoá đó cho $k[n]$ và không

tính tới $k[n]$ nữa (Hình 33). Còn lại dãy khoá $k[1..n-1]$ tuy không còn là biểu diễn của một đồng nữa nhưng nó lại biểu diễn cây nhị phân hoàn chỉnh mà hai nhánh cây ở nút thứ 2 và nút thứ 3 (hai nút con của nút 1) đã là đồng rồi. Vậy chỉ cần vun một lần, ta lại được một đồng, đảo giá trị $k[1]$ cho $k[n-1]$ và tiếp tục cho tới khi đồng chỉ còn lại 1 nút (Hình 34).

Ví dụ:



Hình 33: Đảo giá trị $k[1]$ cho $k[n]$ và xét phần còn lại



Hình 34: Vun phần còn lại thành đồng rồi lại đảo trị $k[1]$ cho $k[n-1]$

Thuật toán HeapSort có hai thủ tục chính:

- ❖ Thủ tục Adjust(root, endnode) vun cây gốc root thành đồng trong điều kiện hai cây gốc $2.root$ và $2.root + 1$ đã là đồng rồi. Các nút từ $endnode + 1$ tới n đã nằm ở vị trí đúng và không được tính tới nữa.
- ❖ Thủ tục HeapSort mô tả lại quá trình vun đồng và chọn khoá theo ý tưởng trên:

```

procedure HeapSort;
var
  r, i: Integer;

procedure Adjust(root, endnode: Integer); {Vun cây gốc Root thành đồng}
var
  c: Integer;
  Key: TKey; {Biến lưu giá trị khoá ở nút Root}
begin
  Key := k[root];
  while root * 2 ≤ endnode do {Chừng nào root chưa phải là lá}
    begin
      c := Root * 2; {Xét nút con trái của Root, so sánh với giá trị nút con phải, chọn ra nút mang giá trị lớn nhất}
      if (c < endnode) and (k[c] < k[c+1]) then c := c + 1;
      if k[c] ≤ Key then Break; {Cả hai nút con của Root đều mang giá trị ≤ Key thì dừng ngay}
      k[root] := k[c]; root := c; {Chuyển giá trị từ nút con c lên nút cha root và đi xuống xét nút con c}
    end;
  k[root] := Key; {Đặt giá trị Key vào nút root}
end;

begin {Bắt đầu thuật toán HeapSort}
  for r := n div 2 downto 1 do Adjust(r, n); {Vun cây từ dưới lên tạo thành đồng}
  for i := n downto 2 do
    begin
      {Đảo giá trị k[1] và k[i]; {Khoá lớn nhất được chuyển ra cuối dãy}
      Adjust(1, i - 1); {Vun phần còn lại thành đồng}
    end;
end;

```

Về độ phức tạp của thuật toán, ta đã biết rằng cây nhị phân hoàn chỉnh có n nút thì chiều cao của nó là $\lceil \lg(n) \rceil + 1$. Cứ cho là trong trường hợp xấu nhất thủ tục Adjust phải thực hiện tìm đường đi từ nút gốc tới nút lá ở xa nhất thì đường đi tìm được cũng chỉ dài bằng chiều cao của cây nên thời gian thực hiện một lần gọi Adjust là $O(\lg n)$. Từ đó có thể suy ra, trong trường hợp xấu nhất, độ phức tạp của HeapSort cũng chỉ là $O(n \lg n)$. Việc đánh giá thời gian thực hiện trung bình phức tạp hơn, ta chỉ ghi nhận một kết quả đã chứng minh được là độ phức tạp trung bình của HeapSort cũng là $O(n \lg n)$.

8.8. SẮP XẾP BẰNG PHÉP Đếm PHÂN PHỐI (DISTRIBUTION COUNTING)

Có một thuật toán sắp xếp đơn giản cho trường hợp đặc biệt: Dãy khoá $k[1..n]$ là các số nguyên nằm trong khoảng từ 0 tới M ($TKey = 0..M$).

Ta dựng dãy $c[0..M]$ các biến đếm, ở đây $c[V]$ là số lần xuất hiện giá trị V trong dãy khoá:

```

for V := 0 to M do c[V] := 0; {Khởi tạo dãy biến đếm}
for i := 1 to n do c[k[i]] := c[k[i]] + 1;

```

Ví dụ với dãy khoá: 1, 2, 2, 3, 0, 0, 1, 1, 3, 3 ($n = 10$, $M = 3$), sau bước đếm ta có:

$c[0] = 2$; $c[1] = 3$; $c[2] = 2$; $c[3] = 3$.

Dựa vào dãy biến đếm, ta hoàn toàn có thể biết được: sau khi sắp xếp thì giá trị V phải nằm từ vị trí nào tới vị trí nào. Như ví dụ trên thì giá trị 0 phải nằm từ vị trí 1 tới vị trí 2; giá trị 1 phải đứng liên tiếp từ vị trí 3 tới vị trí 5; giá trị 2 đứng ở vị trí 6 và 7 còn giá trị 3 nằm ở ba vị trí cuối 8, 9, 10:

0 0 1 1 2 2 3 3 3

Tức là sau khi sắp xếp:

Giá trị 0 đứng trong đoạn từ vị trí 1 tới vị trí c[0].

Giá trị 1 đứng trong đoạn từ vị trí c[0]+1 tới vị trí c[0]+c[1].

Giá trị 2 đứng trong đoạn từ vị trí c[0]+c[1]+1 tới vị trí c[0]+c[1] c[2].

...

Giá trị v trong đoạn đứng từ vị trí c[0]+... +c[v-1]+1 tới vị trí c[0]+...+ c[v].

...

Để ý vị trí cuối của mỗi đoạn, nếu ta tính lại dãy c như sau:

```
for V := 1 to M do c[V] := c[V-1] + c[V]
```

Thì **c[V]** là vị trí cuối của đoạn chứa giá trị V trong dãy khoá đã sắp xếp.

Muốn dựng lại dãy khoá sắp xếp, ta thêm một dãy khoá phụ x[1..n]. Sau đó duyệt lại dãy khoá k, mỗi khi gặp khoá mang giá trị V ta đưa giá trị đó vào khoá x[c[V]] và giảm c[V] đi 1.

```
for i := n downto 1 do
begin
  V := k[i];
  X[c[V]] := k[i]; c[V] := c[V] - 1;
end;
```

Khi đó dãy khoá x chính là dãy khoá đã được sắp xếp, công việc cuối cùng là gán giá trị dãy khoá x cho dãy khoá k.

```
procedure DistributionCounting; { TKey = 0..M}
var
  c: array[0..M] of Integer; { Dãy biến đếm số lần xuất hiện mỗi giá trị}
  t: TArray; { Dãy khoá phụ}
  i: Integer;
  V: TKey;
begin
  for V := 0 to M do c[V] := 0; { Khởi tạo dãy biến đếm}
  for i := 1 to n do c[k[i]] := c[k[i]] + 1; { Đếm số lần xuất hiện các giá trị}
  for V := 1 to M do c[V] := c[V-1] + c[V]; { Tính vị trí cuối mỗi đoạn}
  for i := n downto 1 do
    begin
      V := k[i];
      t[c[V]] := k[i]; c[V] := c[V] - 1;
    end;
  k := x; { Sao chép giá trị từ dãy khoá x sang dãy khoá k}
end;
```

Rõ ràng độ phức tạp của phép đếm phân phối là $O(M + n)$. Nhược điểm của phép đếm phân phối là khi tập giá trị khoá quá lớn thì cho dù n nhỏ cũng không thể làm được.

Có thể có thắc mắc tại sao trong thao tác dựng dãy khoá t, phép duyệt dãy khoá k theo thứ tự nào thì kết quả sắp xếp cũng vẫn đúng, vậy tại sao ta lại chọn phép duyệt ngược từ dưới lên?.

Để trả lời câu hỏi này, ta phải phân tích thêm một đặc trưng của các thuật toán sắp xếp:

8.9. TÍNH ỔN ĐỊNH CỦA THUẬT TOÁN SẮP XẾP (STABILITY)

Một phương pháp sắp xếp được gọi là **ổn định** nếu nó bảo toàn thứ tự ban đầu của các bản ghi mang khoá bằng nhau trong danh sách. Ví dụ như ban đầu danh sách sinh viên được xếp theo thứ tự tên alphabet, thì khi sắp xếp danh sách sinh viên theo thứ tự giảm dần của điểm thi,

những sinh viên bằng nhau sẽ được dồn về một đoạn trong danh sách và vẫn được giữ nguyên thứ tự tên alphabet.

Hãy xem lại những thuật toán sắp xếp ở trước, trong những thuật toán đó, thuật toán sắp xếp nổi bọt, thuật toán sắp xếp chèn và phép đếm phân phối là những thuật toán sắp xếp ổn định, còn những thuật toán sắp xếp khác (và nói chung những thuật toán sắp xếp đòi hỏi phải đảo giá trị 2 bản ghi ở vị trí bất kỳ) là không ổn định.

Với phép đếm phân phối ở mục trước, ta nhận xét rằng nếu hai bản ghi có khoá sắp xếp bằng nhau thì khi đưa giá trị vào dãy bản ghi phụ, bản ghi nào vào trước sẽ nằm phía sau. Vậy nên ta sẽ đẩy giá trị các bản ghi vào dãy phụ theo thứ tự ngược để giữ được thứ tự tương đối ban đầu.

Nói chung, mọi phương pháp sắp xếp tổng quát cho dù không ổn định thì đều có thể biến đổi để nó trở thành ổn định, phương pháp chung nhất được thể hiện qua ví dụ sau:

Giả sử ta cần sắp xếp các sinh viên trong danh sách theo thứ tự giảm dần của điểm bằng một thuật toán sắp xếp ổn định. Ta thêm cho mỗi sinh viên một khoá Index là thứ tự ban đầu của anh ta trong danh sách. Trong thuật toán sắp xếp được áp dụng, cứ chỗ nào cần so sánh hai sinh viên A và B xem ai phải đứng trước, trước hết ta quan tâm tới điểm số: Nếu điểm của A khác điểm của B thì người nào điểm cao hơn sẽ đứng trước, nếu điểm số bằng nhau thì người nào có Index nhỏ hơn sẽ đứng trước.

Trong một số bài toán, tính ổn định của thuật toán sắp xếp quyết định tới cả tính đúng đắn của toàn thuật toán lớn. Chính tính “nhanh” của QuickSort và tính ổn định của phép đếm phân phối là cơ sở nền tảng cho hai thuật toán sắp xếp cực nhanh trên các dãy khoá số mà ta sẽ trình bày dưới đây.

8.10. THUẬT TOÁN SẮP XẾP BẰNG CƠ SỐ (RADIX SORT)


Bài toán đặt ra là: Cho dãy khoá là các số tự nhiên $k[1..n]$ hãy sắp xếp chúng theo thứ tự không giảm. (Trong trường hợp ta đang xét, TKey là kiểu số tự nhiên)

8.10.1. Sắp xếp cơ số theo kiểu hoán vị các khoá (Radix Exchange Sort)

Hãy xem lại thuật toán QuickSort, tại bước phân đoạn nó phân đoạn đang xét thành hai đoạn thoả mãn mỗi khoá trong đoạn đầu \leq mọi khoá trong đoạn sau và thực hiện tương tự trên hai đoạn mới tạo ra, việc phân đoạn được tiến hành với sự so sánh các khoá với giá trị một khoá chốt.

Đối với các số nguyên thì ta có thể coi mỗi số nguyên là một dãy z bit đánh số từ bit 0 (bit ở hàng đơn vị) tới bit $z - 1$ (bit cao nhất).

Ví dụ:



Hình 35: Đánh số các bit

Vậy thì tại bước phân đoạn dãy khoá từ $k[1]$ tới $k[n]$, ta có thể đưa những khoá có bit cao nhất là 0 về đầu dãy, những khoá có bit cao nhất là 1 về cuối dãy. Để thấy rằng những khoá bắt đầu bằng bit 0 sẽ phải nhỏ hơn những khoá bắt đầu bằng bit 1. Tiếp tục quá trình phân đoạn với hai đoạn dãy khoá: Đoạn gồm các khoá có bit cao nhất là 0 và đoạn gồm các khoá có bit cao nhất là 1. Với những khoá thuộc cùng một đoạn thì có bit cao nhất giống nhau, nên ta có thể áp dụng quá trình phân đoạn tương tự trên theo bit thứ $z - 2$ và cứ tiếp tục như vậy ...

Quá trình phân đoạn kết thúc nếu như đoạn đang xét là rỗng hay ta đã tiến hành phân đoạn đến tận bit đơn vị, tức là tất cả các khoá thuộc một trong hai đoạn mới tạo ra đều có bit đơn vị bằng nhau (điều này đồng nghĩa với sự bằng nhau ở tất cả những bit khác, tức là bằng nhau về giá trị khoá).

Ví dụ:

Xét dãy khoá: 1, 3, 7, 6, 5, 2, 3, 4, 4, 5, 6, 7. Tương ứng với các dãy 3 bit:

001	011	111	110	101	010	011	100	100	101	110	111
-----	-----	-----	-----	-----	-----	-----	-----	-----	-----	-----	-----

Trước hết ta chia đoạn dựa vào bit 2 (bit cao nhất):

001	011	011	010	101	110	111	100	100	101	110	111
-----	-----	-----	-----	-----	-----	-----	-----	-----	-----	-----	-----

Sau đó chia tiếp hai đoạn tạo ra dựa vào bit 1:

001	011	011	010	101	101	100	100	111	110	110	111
-----	-----	-----	-----	-----	-----	-----	-----	-----	-----	-----	-----

Cuối cùng, chia tiếp những đoạn tạo ra dựa vào bit 0:

001	010	011	011	100	100	101	101	110	110	111	111
-----	-----	-----	-----	-----	-----	-----	-----	-----	-----	-----	-----

Ta được dãy khoá tương ứng: 1, 2, 3, 3, 4, 4, 5, 5, 6, 6, 7, 7 là dãy khoá sắp xếp.

Quá trình chia đoạn dựa vào bit b có thể chia thành một đoạn rỗng và một đoạn gồm toàn bộ các khoá còn lại, nhưng việc chia đoạn không bao giờ bị rơi vào quá trình đệ quy vô hạn bởi những lần đệ quy tiếp theo sẽ phân đoạn dựa vào bit $b - 1$, $b - 2$... và sẽ phải dừng lại khi xét tới bit 0. Công việc còn lại là cố gắng hiểu đoạn chương trình sau và phân tích xem tại sao nó hoạt động đúng:

```

procedure RadixExchangeSort;
var
  z: Integer; {Độ dài dãy bit biểu diễn mỗi khoá}

procedure Partition(L, H, b: Integer); {Phân đoạn [L, H] dựa vào bit b}
var
  i, j: Integer;
begin
  if L ≥ H then Exit;
  i := L; j := H;
  repeat
    {Hai vòng lặp trong dưới đây luôn cầm canh i < j}
    while (i < j) and (Bit b của k[i] = 0) do i := i + 1; {Tìm khoá có bit b = 1 từ đầu đoạn}
    while (i < j) and (Bit b của k[j] = 1) do j := j - 1; {Tìm khoá có bit b = 0 từ cuối đoạn}
    {Đảo giá trị k[i] cho k[j];}
  until i = j;
  if <Bit b của k[j] = 0> then j := j + 1; {j là điểm bắt đầu của đoạn có bit b là 1}
  if b > 0 then {Chưa xét tới bit đơn vị}
    begin
      Partition(L, j - 1, b - 1); Partition(j, R, b - 1);
    end;
  end;
end;

begin
  {Dựa vào giá trị lớn nhất của dãy khoá, xác định z là độ dài dãy bit biểu diễn mỗi khoá};
  Partition(1, n, z - 1);
end;

```

Với Radix Exchange Sort, ta hoàn toàn có thể làm trên hệ cơ số R khác chứ không nhất thiết phải làm trên hệ nhị phân (ý tưởng cũng tương tự như trên), tuy nhiên quá trình phân đoạn sẽ không phải chia làm 2 mà chia thành R đoạn. Về độ phức tạp của thuật toán, ta thấy để phân đoạn bằng một bit thì thời gian sẽ là $C \cdot n$ để chia tất cả các đoạn cần chia bằng bit đó (C là hằng số). Vậy tổng thời gian phân đoạn bằng z bit sẽ là $C \cdot n \cdot z$. Trong trường hợp xấu nhất, độ phức tạp của Radix Exchange Sort là $O(n \cdot z)$. Và độ phức tạp trung bình của Radix Exchange Sort là $O(n \cdot \min(z, \lg n))$.

Nói chung, Radix Exchange Sort cài đặt như trên chỉ thể hiện tốc độ tối đa trên các hệ thống cho phép xử lý trực tiếp trên các bit: Hệ thống phải cho phép lấy một bit ra dễ dàng và thao tác với thời gian nhanh hơn hẳn so với thao tác trên BYTE, WORD, DWORD, QWORD... Khi đó Radix Exchange Sort sẽ tốt hơn nhiều QuickSort. (Ta thử lập trình sắp xếp các dãy nhị phân độ dài z theo thứ tự từ điển để khảo sát). Trên các máy tính hiện nay chỉ cho phép xử lý trực tiếp trên BYTE (hay WORD, DWORD v.v...), việc tách một bit ra khỏi Byte đó để xử lý lại rất chậm và làm ảnh hưởng không nhỏ tới tốc độ của Radix Exchange Sort. Chính vì vậy, tuy đây là một phương pháp hay, nhưng khi cài đặt cụ thể thì tốc độ cũng chỉ ngang ngửa chứ không thể qua mặt QuickSort được.

8.10.2. Sắp xếp cơ số trực tiếp (Straight Radix Sort)

Ta sẽ trình bày phương pháp sắp xếp cơ số trực tiếp bằng một ví dụ: Sắp xếp dãy khoá:

925	817	821	638	639	744	742	563	570	166
-----	-----	-----	-----	-----	-----	-----	-----	-----	-----

Trước hết, ta sắp xếp dãy khoá này theo thứ tự tăng dần của chữ số hàng đơn vị bằng một thuật toán sắp xếp khác, được dãy khoá:

570	821	742	563	744	925	166	817	638	639
-----	-----	-----	-----	-----	-----	-----	-----	-----	-----

Sau đó, ta sắp xếp dãy khoá mới tạo thành theo thứ tự tăng dần của chữ số hàng chục bằng một thuật toán sắp xếp **ổn định**, được dãy khoá:

817	821	925	638	639	742	744	563	166	570
-----	-----	-----	-----	-----	-----	-----	-----	-----	-----

Vì thuật toán sắp xếp ta sử dụng là ổn định, nên nếu hai khoá có chữ số hàng chục giống nhau thì khoá nào có chữ số hàng đơn vị nhỏ hơn sẽ đứng trước. Nói như vậy có nghĩa là dãy khoá thu được sẽ có thứ tự tăng dần về giá trị tạo thành từ hai chữ số cuối.

Cuối cùng, ta sắp xếp lại dãy khoá theo thứ tự tăng dần của chữ số hàng trăm cũng bằng một thuật toán sắp xếp ổn định, thu được dãy khoá:

166	563	570	638	639	742	744	817	821	925
-----	-----	-----	-----	-----	-----	-----	-----	-----	-----

Lập luận tương tự như trên dựa vào tính ổn định của phép sắp xếp, dãy khoá thu được sẽ có thứ tự tăng dần về giá trị tạo thành bởi cả ba chữ số, đó là dãy khoá đã sắp.

Nhận xét:

Ta hoàn toàn có thể coi số chữ số của mỗi khoá là bằng nhau, như ví dụ trên nếu có số 15 trong dãy khoá thì ta có thể coi nó là 015.

Cũng từ ví dụ, ta có thể thấy rằng số lượt thao tác sắp xếp phải áp dụng đúng bằng số chữ số tạo thành một khoá. Với một hệ cơ số lớn, biểu diễn một giá trị khoá sẽ phải dùng ít chữ số hơn. Ví dụ số 12345 trong hệ thập phân phải dùng tới 5 chữ số, còn trong hệ cơ số 1000 chỉ cần dùng 2 chữ số AB mà thôi, ở đây A là chữ số mang giá trị 12 còn B là chữ số mang giá trị 345.

Tốc độ của sắp xếp cơ số trực tiếp phụ thuộc rất nhiều vào thuật toán sắp xếp ổn định tại mỗi bước. Không có một lựa chọn nào khác tốt hơn phép đếm phân phối. Tuy nhiên, phép đếm phân phối có thể không cài đặt được hoặc kém hiệu quả nếu như tập giá trị khoá quá rộng, không cho phép dựng ra dãy các biến đếm hoặc phải sử dụng dãy biến đếm quá dài (Điều này xảy ra nếu chọn hệ cơ số quá lớn).

Một lựa chọn khôn ngoan là nên chọn hệ cơ số thích hợp cho từng trường hợp cụ thể để dung hoà tới mức tối ưu nhất ba mục tiêu:

- ❖ Việc lấy ra một chữ số của một số được thực hiện dễ dàng
- ❖ Sử dụng ít lần gọi phép đếm phân phối.
- ❖ Phép đếm phân phối thực hiện nhanh

```

procedure StraightRadixSort;
const
  radix = ...; {Tuỳ chọn hệ cơ số radix cho hợp lý}
var
  t: TArray; {Dãy khoá phụ}
  p: Integer;
  nDigit: Integer; {Số chữ số cho một khoá, đánh số từ chữ số thứ 0 là hàng đơn vị đến chữ số thứ nDigit - 1}
  Flag: Boolean; {Flag = True thì sắp dãy k, ghi kết quả vào dãy t; Flag = False thì sắp dãy t, ghi kq vào k}

  function GetDigit(Num: TKey; p: Integer): Integer; {Lấy chữ số thứ p của số Num (0≤p<nDigit)}
begin
  GetDigit := Num div radixp mod radix; {Trường hợp cụ thể có thể có cách viết tốt hơn}
end;

{Sắp xếp ổn định dãy số x theo thứ tự tăng dần của chữ số thứ p, kết quả sắp xếp được chia vào dãy số y}
procedure DCount(var x, y: TArray; p: Integer); {Thuật toán đếm phân phối, sắp từ x sang y}
var
  c: array[0..radix - 1] of Integer; {c[d] là số lần xuất hiện chữ số d tại vị trí p}
  i, d: Integer;
begin
  for d := 0 to radix - 1 do c[d] := 0;
  for i := 1 to n do
    begin
      d := GetDigit(x[i], p); c[d] := c[d] + 1;
    end;
  for d := 1 to radix - 1 do c[d] := c[d-1] + c[d]; {các c[d] trở thành các mốc cuối đoạn}
  for i := n downto 1 do {Điền giá trị vào dãy y}
    begin
      d := GetDigit(x[i], p);
      y[c[d]] := x[i]; c[d] := c[d] - 1;
    end;
  end;

begin {Thuật toán sắp xếp cơ số trực tiếp}
  {Dựa vào giá trị lớn nhất trong dãy khoá, xác định nDigit là số chữ số phải dùng cho mỗi khoá trong hệ radix};
  Flag := True;
  for p := 0 to nDigit - 1 do {Xét từ chữ số hàng đơn vị lên, sắp xếp ổn định theo chữ số thứ p}
    begin
      if Flag then DCount(k, t, p) else DCount(t, k, p);
      Flag := not Flag; {Đảo cờ, dùng k tính t rồi lại dùng t tính k ...}
    end;
  if not Flag then k := t; {Nếu kết quả cuối cùng đang ở trong t thì sao chép giá trị từ t sang k}
end;

```

Xét phép đếm phân phối, ta đã biết độ phức tạp của nó là $O(\max(\text{radix}, n))$. Mà radix là một hằng số tự ta chọn từ trước, nên khi n lớn, độ phức tạp của phép đếm phân phối là $O(n)$. Thuật toán sử dụng nDigit lần phép đếm phân phối nên có thể thấy độ phức tạp của thuật toán là $O(n \cdot nDigit)$ bất kể dữ liệu đầu vào.

Ta có thể coi sắp xếp cơ số trực tiếp là một mở rộng của phép đếm phân phối, khi dãy số chỉ toàn các số có 1 chữ số (trong hệ radix) thì đó chính là phép đếm phân phối. Sự khác biệt ở đây là: Sắp xếp cơ số trực tiếp có thể thực hiện với các khoá mang giá trị lớn; còn phép đếm phân phối chỉ có thể làm trong trường hợp các khoá mang giá trị nhỏ, bởi nó cần một lượng bộ nhớ đủ rộng để giăng ra dãy biến đếm số lần xuất hiện cho từng giá trị.

8.11. THUẬT TOÁN SẮP XẾP TRỘN (MERGESORT)

Thuật toán sắp xếp trộn (MergeSort hay Collation Sort) là một trong những thuật toán sắp xếp cổ điển nhất, được đề xuất bởi J.von Neumann năm 1945. Cho tới nay, người ta vẫn coi MergeSort là một thuật toán sắp xếp ngoài mực, được đưa vào giảng dạy rộng rãi và được tích hợp trong nhiều phần mềm thương mại.

8.11.1. Phép trộn 2 đường


Phép trộn 2 đường là phép hợp nhất hai dãy khoá **đã sắp xếp** để ghép lại thành một dãy khoá có kích thước bằng tổng kích thước của hai dãy khoá ban đầu và dãy khoá tạo thành cũng có thứ tự sắp xếp. Nguyên tắc thực hiện của nó khá đơn giản: so sánh hai khoá đứng đầu hai dãy, chọn ra khoá nhỏ nhất và đưa nó vào miền sắp xếp (một dãy khoá phụ có kích thước bằng tổng kích thước hai dãy khoá ban đầu) ở vị trí thích hợp. Sau đó, khoá này bị loại ra khỏi dãy khoá chứa nó. Quá trình tiếp tục cho tới khi một trong hai dãy khoá đã cạn, khi đó chỉ cần chuyển toàn bộ dãy khoá còn lại ra miền sắp xếp là xong.

Ví dụ: Với hai dãy khoá: (1, 3, 10, 11) và (2, 4, 9)

Dãy 1	Dãy 2	Khoá nhỏ nhất trong 2 dãy	Miền sắp xếp
(1, 3, 10, 11)	(2, 4, 9)	1	(1)
(3, 10, 11)	(2, 4, 9)	2	(1, 2)
(3, 10, 11)	(4, 9)	3	(1, 2, 3)
(10, 11)	(4, 9)	4	(1, 2, 3, 4)
(10, 11)	(9)	9	(1, 2, 3, 4, 9)
(10, 11)	Ø	Dãy 2 là Ø, đưa nốt dãy 1 vào miền sắp xếp	(1, 2, 3, 4, 9, 10, 11)

8.11.2. Sắp xếp bằng trộn 2 đường trực tiếp

Ta có thể coi mỗi khoá trong dãy khoá $k[1..n]$ là một mạch với độ dài 1, dĩ nhiên các mạch độ dài 1 có thể coi là đã được sắp. Nếu trộn hai mạch liên tiếp lại thành một mạch có độ dài 2, ta lại được dãy gồm các mạch đã được sắp. Cứ tiếp tục như vậy, số mạch trong dãy sẽ giảm dần sau mỗi lần trộn (Hình 36)



Hình 36: Thuật toán sắp xếp trộn

Để tiến hành thuật toán sắp xếp trộn hai đường trực tiếp, ta viết các thủ tục:

Thủ tục Merge(var x, y: TArray; a, b, c: Integer); thủ tục này trộn mảnh x[a..b] với mảnh x[b+1..c] để được mảnh y[a..c].

Thủ tục MergeByLength(var x, y: TArray; len: Integer); thủ tục này trộn lần lượt các cặp mảnh theo thứ tự:

Trộn mảnh x[1..len] và x[len+1..2len] thành mảnh y[1..2len].

Trộn mảnh x[2len+1..3len] và x[3len+1..4len] thành mảnh y[2len+1..4len].

...

Lưu ý rằng đến cuối cùng ta có thể gặp hai trường hợp: Hoặc còn lại hai mảnh mà mảnh thứ hai có độ dài $< len$. Hoặc chỉ còn lại một mảnh. Trường hợp thứ nhất ta phải quản lý chính xác các chỉ số để thực hiện phép trộn, còn trường hợp thứ hai thì không được quên thao tác đưa thẳng mảnh duy nhất còn lại sang dãy y.

Cuối cùng là thủ tục MergeSort, thủ tục này cần một dãy khoá phụ t[1..n]. Trước hết ta gọi MergeByLength(k, t, 1) để trộn hai khoá liên tiếp của k thành một mảnh trong t, sau đó lại gọi MergeByLength(t, k, 2) để trộn hai mảnh liên tiếp trong t thành một mảnh trong k, rồi lại gọi MergeByLength(k, t, 4) để trộn hai mảnh liên tiếp trong k thành một mảnh trong t ... Như vậy k và t được sử dụng với vai trò luân phiên: một dãy chứa các mảnh và một dãy dùng để trộn các cặp mảnh liên tiếp để được mảnh lớn hơn.

```

procedure MergeSort;
var
  t: TArray; {Dãy khoá phụ}
  len: Integer;
  Flag: Boolean; {Flag = True: trộn các mảnh trong k vào t; Flag = False: trộn các mảnh trong t vào k}

procedure Merge(var X, Y: TArray; a, b, c: Integer);{Trộn X[a..b] và X[b+1..c]}
var
  i, j, p: Integer;
begin
  {Chi số p chạy trong miền sắp xếp, i chạy theo mảnh thứ nhất, j chạy theo mảnh thứ hai}
  p := a; i := a; j := b + 1;
  while (i ≤ b) and (j ≤ c) then {Chừng nào cả hai mảnh đều chưa xét hết}
    begin
      if X[i] ≤ X[j] then {So sánh hai khoá nhỏ nhất trong hai mảnh mà chưa bị đưa vào miền sắp xếp}
        begin
          Y[p] := X[i]; i := i + 1; {Đưa x[i] vào miền sắp xếp và cho i chạy}
        end
      else
        begin
          Y[p] := X[j]; j := j + 1; {Đưa x[j] vào miền sắp xếp và cho j chạy}
        end;
      p := p + 1;
    end;
  if i ≤ b then Y[p..c] := X[i..b] {Mảnh 2 hết trước, Đưa phần cuối của mảnh 1 vào miền sắp xếp}
  else Y[p..c] := X[j..c]; {Mảnh 1 hết trước, Đưa phần cuối của mảnh 2 vào miền sắp xếp}
end;

procedure MergeByLength(var X, Y: TArray; len: Integer);
begin
  a := 1; b := len; c := 2 * len;
  while c ≤ n do {Trộn hai mảnh x[a..b] và x[b+1..c] đều có độ dài len}
    begin
      Merge(X, Y, a, b, c);
      a := a + 2 * len; b := b + 2 * len; c := c + 2 * len; {Dịch các chỉ số a, b, c về sau 2.len vị trí}
    end;
  if b < n then Merge(X, Y, a, b, n) {Còn lại hai mảnh mà mảnh thứ hai có độ dài ngắn hơn len}
  else
    if a ≤ n then Y[a..n] := X[a..n] {Còn lại một mảnh thì đưa thẳng mảnh đó sang miền y}
  end;

begin {Thuật toán sắp xếp trộn}
  Flag := True;
  len := 1;
  while len < n do
    begin
      if Flag then MergeByLength(k, t, len) else MergeByLength(t, k, len);
      len := len * 2;
      Flag := not Flag; {Đảo cờ để luôn phiên vai trò của k và t}
    end;
  if not Flag then k := t; {Nếu kết quả cuối cùng đang nằm trong t thì sao chép kết quả vào k}
end;

```

Về độ phức tạp của thuật toán, ta thấy rằng trong thủ tục Merge, phép toán tích cực là thao tác đưa một khoá vào miền sắp xếp. Mỗi lần gọi thủ tục MergeByLength, tất cả các khoá trong dãy khoá được chuyển hoàn toàn sang miền sắp xếp, nên độ phức tạp của thủ tục MergeByLength là $O(n)$. Thủ tục MergeSort có vòng lặp thực hiện không quá $\lceil \lg n \rceil$ lần gọi MergeByLength bởi biến len sẽ được tăng theo cấp số nhân công bội 2. Từ đó suy ra độ phức tạp của MergeSort là $O(n \lg n)$ bất chấp trạng thái dữ liệu vào.

Cùng là những thuật toán sắp xếp tổng quát với độ phức tạp trung bình như nhau, nhưng không giống như QuickSort hay HeapSort, MergeSort có tính **ổn định**. Nhược điểm của MergeSort là nó phải dùng thêm một vùng nhớ để chứa dãy khoá phụ có kích thước bằng dãy khoá ban đầu.

Người ta còn có thể lợi dụng được trạng thái dữ liệu vào để khiến MergeSort chạy nhanh hơn: ngay từ đầu, ta không coi mỗi khoá của dãy khoá là một mảnh mà coi những đoạn đã được sắp trong dãy khoá là một mảnh. Bởi một dãy khoá bất kỳ có thể coi là gồm các mảnh đã sắp xếp nằm liên tiếp nhau. Khi đó người ta gọi phương pháp này là phương pháp **trộn hai đường tự nhiên**.

Tổng quát hơn nữa, thay vì phép trộn hai mảnh, người ta có thể sử dụng phép trộn k mảnh, khi đó ta được thuật toán sắp xếp trộn k đường.

8.12. CÀI ĐẶT

Ta sẽ cài đặt tất cả các thuật toán sắp xếp nêu trên, với dữ liệu vào được đặt trong file văn bản SORT.INP chứa không nhiều hơn 10^6 khoá và giá trị mỗi khoá là số tự nhiên không quá 10^6 . Kết quả được ghi ra file văn bản SORT.OUT chứa dãy khoá được sắp, mỗi khoá trên một dòng.

SORT.INP	SORT.OUT
1 4 3 2 5	1
7 9 8	2
10 6	3
	4
	5
	6
	7
	8
	9
	10

Chương trình có giao diện dưới dạng menu, mỗi chức năng tương ứng với một thuật toán sắp xếp. Tại mỗi thuật toán sắp xếp, ta thêm một vài lệnh đo thời gian thực tế của nó (chỉ đo thời gian thực hiện giải thuật, không tính thời gian nhập liệu và in kết quả).

Ở thuật toán Radix Exchange Sort, ta chọn hệ nhị phân. Ở thuật toán Straight Radix Sort, ta sử dụng hệ cơ số 256, khi đó nếu một giá trị số tự nhiên x biểu diễn bằng $d + 1$ chữ số trong hệ 256: $x = \overline{x_d \dots x_1 x_0}_{(256)}$ thì $x_p = x \text{ div } 256^p \text{ mod } 256 = (x \text{ shr } (p \text{ shl } 3)) \text{ and } \FF ($1 \leq p \leq d$).

P_2_08_1.PAS * Các thuật toán sắp xếp

```
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)  
program SortingAlgorithmsDemo;  
uses crt, dos;  
const  
  InputFile = 'SORT.INP';  
  OutputFile = 'SORT.OUT';  
  max = 1000000;  
  maxV = 1000000;  
  BitCount = 64;  
  nMenu = 13;  
  SMenu: array[1..nMenu] of String =  
  (
```

```
'D. Display Input',
'1. SelectionSort',
'2. BubbleSort',
'3. InsertionSort',
'4. InsertionSort with binary searching',
'5. ShellSort',
'6. QuickSort',
'7. HeapSort',
'8. Distribution Counting',
'9. Radix Exchange Sort',
'A. Straight Radix Sort',
'B. MergeSort',
'E. Exit'

);

type
  TArr = array[1..max] of Integer;
  TCount = array[0..maxV] of Integer;
var
  k, t: TArr;
  c: TCount;
  n, MinV, SupV: Integer;
  selected: Integer;
  StTime: Extended;

function GetcurrentTime: Extended;
var
  h, m, s, s100: Word;
begin
  GetTime(h, m, s, s100);
  GetcurrentTime := (h * 3600 + m * 60 + s) + s100 / 100;
end;

procedure Enter;
var
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  n := 0;
  MinV := High(Integer); SupV := 0;
  while not SeekEof(f) do
    begin
      Inc(n); Read(f, k[n]);
      if k[n] < MinV then MinV := k[n];
      if k[n] > SupV then SupV := k[n];
    end;
  Close(f);
  Inc(SupV);
  StTime := GetcurrentTime;
end;

procedure PrintInput;
var
  i: Integer;
begin
  Enter;
  for i := 1 to n do Write(k[i]:8);
  Write('Press any key to return to menu... ');
  ReadKey;
end;

procedure PrintResult;
var
  f: Text;
```

```

i: Integer;
ch: Char;
begin
  Writeln('Running Time = ', GetCurrentTime - StTime:1:4, ' (s)');
  Assign(f, OutputFile); Rewrite(f);
  for i := 1 to n do Writeln(f, k[i]);
  Close(f);
  Write('Press <P> to print Output, another key to return to menu... ');
  ch := ReadKey; Writeln(ch);
  if Upcase(ch) = 'P' then
    begin
      for i := 1 to n do Write(k[i]:8);
      Writeln;
      Write('Press any key to return to menu... ');
      ReadKey;
    end;
  end;

procedure Swap(var x, y: Integer);
var
  t: Integer;
begin
  t := x; x := y; y := t;
end;

{----- Sorting Algorithms -----}
{ SelectionSort }

procedure SelectionSort;
var
  i, j, jmin: Integer;
begin
  Enter;
  for i := 1 to n - 1 do
    begin
      jmin := i;
      for j := i + 1 to n do
        if k[j] < k[jmin] then jmin := j;
      if jmin <> i then Swap(k[i], k[jmin]);
    end;
  PrintResult;
end;

{ BubbleSort }

procedure BubbleSort;
var
  i, j: Integer;
begin
  Enter;
  for i := 2 to n do
    for j := n downto i do
      if k[j - 1] > k[j] then Swap(k[j - 1], k[j]);
  PrintResult;
end;

{ InsertionSort }

procedure InsertionSort;
var
  i, j, tmp: Integer;
begin
  Enter;

```

```

for i := 2 to n do
begin
  tmp := k[i]; j := i - 1;
  while (j > 0) and (tmp < k[j]) do
    begin
      k[j + 1] := k[j];
      Dec(j);
    end;
  k[j + 1] := tmp;
end;
PrintResult;
end;

{ InsertionSort with Binary searching }

procedure AdvancedInsertionSort;
var
  i, inf, sup, median, tmp: Integer;
begin
  Enter;
  for i := 2 to n do
    begin
      tmp := k[i];
      inf := 1; sup := i - 1;
      repeat
        median := (inf + sup) shr 1;
        if tmp < k[median] then sup := median - 1
        else inf := median + 1;
      until inf > sup;
      Move(k[inf], k[inf + 1], (i - inf) * SizeOf(k[1]));
      k[inf] := tmp;
    end;
  PrintResult;
end;

{ ShellSort }

procedure ShellSort;
var
  tmp: Integer;
  i, j, h: Integer;
begin
  Enter;
  h := n shr 1;
  while h <> 0 do
    begin
      for i := h + 1 to n do
        begin
          tmp := k[i]; j := i - h;
          while (j > 0) and (k[j] > tmp) do
            begin
              k[j + h] := k[j];
              j := j - h;
            end;
          k[j + h] := tmp;
        end;
      h := h shr 1;
    end;
  PrintResult;
end;

{ QuickSort }

```

```

procedure QuickSort;

procedure Partition(L, H: Integer);
var
  i, j: Integer;
  Pivot: Integer;
begin
  if L >= H then Exit;
  Pivot := k[L + Random(H - L + 1)];
  i := L; j := H;
  repeat
    while k[i] < Pivot do Inc(i);
    while k[j] > Pivot do Dec(j);
    if i <= j then
      begin
        if i < j then Swap(k[i], k[j]);
        Inc(i); Dec(j);
      end;
    until i > j;
  Partition(L, j); Partition(i, H);
end;

begin
  Enter;
  Partition(1, n);
  PrintResult;
end;

{ HeapSort }

procedure HeapSort;
var
  r, i: Integer;

procedure Adjust(root, endnode: Integer);
var
  key, c: Integer;
begin
  key := k[root];
  while root shr 1 <= endnode do
    begin
      c := root shr 1;
      if (c < endnode) and (k[c] < k[c + 1]) then Inc(c);
      if k[c] <= key then Break;
      k[root] := k[c]; root := c;
    end;
  k[root] := key;
end;

begin
  Enter;
  for r := n shr 1 downto 1 do Adjust(r, n);
  for i := n downto 2 do
    begin
      Swap(k[1], k[i]);
      Adjust(1, i - 1);
    end;
  PrintResult;
end;

{ Distribution Counting }

procedure DistributionCounting;

```

```

var
  i, V: Integer;
begin
  Enter;
  FillChar(c, SizeOf(c), 0);
  for i := 1 to n do Inc(c[k[i]]);
  for V := MinV + 1 to SupV - 1 do c[V] := c[V - 1] + c[V];
  for i := n downto 1 do
    begin
      V := k[i];
      t[c[V]] := k[i];
      Dec(c[V]);
    end;
  k := t;
  PrintResult;
end;

{ Radix Exchange Sort }

procedure RadixExchangeSort;
var
  MaskBit: array[0..BitCount - 1] of Integer;
  i, maxbit: Integer;

procedure Partition(L, H, BIndex: Integer);
var
  i, j, Mask: Integer;
begin
  if L >= H then Exit;
  i := L; j := H; Mask := MaskBit[BIndex];
  repeat
    while (i < j) and (k[i] and Mask = 0) do Inc(i);
    while (i < j) and (k[j] and Mask <> 0) do Dec(j);
    Swap(k[i], k[j]);
  until i = j;
  if k[j] and Mask = 0 then Inc(j);
  if BIndex > 0 then
    begin
      Partition(L, j - 1, BIndex - 1); Partition(j, H, BIndex - 1);
    end;
end;

begin
  Enter;
  maxbit := Trunc(Ln(SupV) / Ln(2));
  for i := 0 to maxbit do MaskBit[i] := 1 shl i;
  Partition(1, n, maxbit);
  PrintResult;
end;

{ Straight Radix Sort}

procedure StraightRadixSort;
const
  Radix = 256;
var
  p, maxDigit: Integer;
  Flag: Boolean;

  function GetDigit(key, p: Integer): Integer;
begin
  GetDigit := (key shr (p shl 3)) and $FF;
end;

```

```

procedure DCount(var x, y: TArr; p: Integer);
var
  c: array[0..Radix - 1] of Integer;
  i, d: Integer;
begin
  FillChar(c, SizeOf(c), 0);
  for i := 1 to n do
    begin
      d := GetDigit(x[i], p); Inc(c[d]);
    end;
  for d := 1 to Radix - 1 do c[d] := c[d - 1] + c[d];
  for i := n downto 1 do
    begin
      d := GetDigit(x[i], p);
      y[c[d]] := x[i];
      Dec(c[d]);
    end;
end;

begin
  Enter;
  MaxDigit := Trunc(Ln(SupV) / Ln(Radix));
  Flag := True;
  for p := 0 to MaxDigit do
    begin
      if Flag then DCount(k, t, p)
      else DCount(t, k, p);
      Flag := not Flag;
    end;
  if not Flag then k := t;
  PrintResult;
end;

{ MergeSort }

procedure MergeSort;
var
  Flag: Boolean;
  len: Integer;

procedure Merge(var Source, Dest: TArr; a, b, c: Integer);
var
  i, j, p: Integer;
begin
  p := a; i := a; j := b + 1;
  while (i <= b) and (j <= c) do
    begin
      if Source[i] <= Source[j] then
        begin
          Dest[p] := Source[i]; Inc(i);
        end
      else
        begin
          Dest[p] := Source[j]; Inc(j);
        end;
      Inc(p);
    end;
  if i <= b then
    Move(Source[i], Dest[p], (b - i + 1) * SizeOf(Source[1]))
  else
    Move(Source[j], Dest[p], (c - j + 1) * SizeOf(Source[1]));
end;

```

```

procedure MergeByLength(var Source, Dest: TArr; len: Integer);
var
  a, b, c: Integer;
begin
  a := 1; b := len; c := len shl 1;
  while c <= n do
    begin
      Merge(Source, Dest, a, b, c);
      a := a + len shl 1; b := b + len shl 1; c := c + len shl 1;
    end;
  if b < n then Merge(Source, Dest, a, b, n)
  else
    if a <= n then
      Move(Source[a], Dest[a], (n - a + 1) * SizeOf(Source[1]));
  end;

begin
  Enter;
  len := 1; Flag := True;
  FillChar(t, SizeOf(t), 0);
  while len < n do
    begin
      if Flag then MergeByLength(k, t, len)
      else MergeByLength(t, k, len);
      len := len shl 1;
      Flag := not Flag;
    end;
  if not Flag then k := t;
  PrintResult;
end;
{----- End of Sorting Algorithms -----}

function MenuSelect: Integer;
var
  i: Integer;
  ch: Char;
begin
  Clrscr;
  Writeln('Sorting Algorithms Demos; Input: SORT.INP; Output: SORT.OUT');
  for i := 1 to nMenu do Writeln(' ', SMenu[i]);
  Write('Enter your choice: ');
  ch := Upcase(ReadKey);
  Writeln(ch);
  for i := 1 to nMenu do
    if SMenu[i][1] = ch then
      begin
        MenuSelect := i;
        Exit;
      end;
  MenuSelect := 0;
end;

begin
  repeat
    selected := MenuSelect;
    if not (Selected in [1..nMenu]) then Continue;
    Writeln(SMenu[selected]);
    case selected of
      1 : PrintInput;
      2 : SelectionSort;
      3 : BubbleSort;
      4 : InsertionSort;
    end;
  until false;
end.

```

```

5 : AdvancedInsertionSort;
6 : ShellSort;
7 : QuickSort;
8 : HeapSort;
9 : DistributionCounting;
10: RadixExchangeSort;
11: StraightRadixSort;
12: MergeSort;
13: Halt;
end;
until False;
end.

```


8.13. ĐÁNH GIÁ, NHẬN XÉT

Những con số về thời gian và tốc độ chương trình đo được là qua thử nghiệm trên một bộ dữ liệu cụ thể, với một máy tính cụ thể và một công cụ lập trình cụ thể. Với bộ dữ liệu khác, máy tính và công cụ lập trình khác, kết quả có thể khác. Tôi đã viết lại chương trình này trên Borland Delphi 7 để đưa vào một số cải tiến:

- ❖ Thiết kế dựa trên kiến trúc đa luồng (MultiThreads) cho phép chạy đồng thời hai hay nhiều thuật toán sắp xếp để so sánh tốc độ, bên cạnh đó vẫn có thể chạy tuần tự các thuật toán sắp xếp để đo thời gian thực hiện chính xác của chúng.
- ❖ Quá trình sắp xếp được hiển thị trực quan trên màn hình.
- ❖ Bỏ đi thuật toán sắp xếp kiểu chèn dạng nguyên thuỷ, chỉ giữ lại thuật toán sắp xếp kiểu chèn dùng tìm kiếm nhị phân
- ❖ Thuật toán ShellSort được viết lại, dùng các giá trị trong dãy Pratt: 1, 2, 3, 4, 6, 8, 9, 12, 16, ..., $2^i 3^j$, ... làm độ dài bước.

Thử nghiệm cả hai chương trình, một trên Free Pascal 1.0.10 và một trên Delphi 7, nhìn chung tốc độ sắp xếp của chương trình viết trên Delphi nhanh hơn nhiều so với chương trình trên FPK, tuy nhiên khi so sánh tốc độ tương đối giữa các thuật toán vẫn có nhiều khác biệt giữa hai chương trình. Có một số thuật toán thực hiện nhanh bất ngờ so với dự đoán và cũng có một số thuật toán thực hiện chậm hơn hẳn so với đánh giá lý thuyết. Có thể có gắng giải thích qua hiệu ứng của bộ nhớ Cache và cách thức truy cập bộ nhớ nhưng điều này hơi phức tạp và cũng không thực sự cần thiết. Ta chỉ rút ra kinh nghiệm rằng tốc độ thực thi của một thuật toán phụ thuộc rất nhiều vào phần cứng máy tính và chương trình dịch.

Hình 37 là giao diện của chương trình viết trên Delphi, bạn có thể tham khảo mã nguồn kèm theo. Có một điều phải lưu ý là để chương trình không bị ảnh hưởng bởi các phần mềm khác đang chạy, khi khởi động các threads, bàn phím, chuột và tất cả các phần mềm khác sẽ bị treo tạm thời đến khi các threads thực hiện xong. Vì vậy không nên chạy các thuật toán sắp xếp chậm với dữ liệu lớn vì sẽ không thể đợi đến khi các threads kết thúc và sẽ phải tắt máy khởi động lại.



Hình 37: Máy Pentium 4, 3.2GHz, 2GB RAM tỏ ra chậm chạp khi sắp xếp 10^8 khoá $\in [0..7.10^7]$ cho dù những thuật toán sắp xếp tốt nhất đã được áp dụng

Cùng một mục đích sắp xếp như nhau, nhưng có nhiều phương pháp giải quyết khác nhau. Nếu chỉ dựa vào thời gian đo được trong một ví dụ cụ thể mà đánh giá thuật toán này tốt hơn thuật toán kia về mọi mặt là điều không nên. Việc chọn một thuật toán sắp xếp thích hợp cho phù hợp với từng yêu cầu, từng điều kiện cụ thể là kỹ năng của người lập trình.

Những thuật toán có độ phức tạp $O(n^2)$ thì chỉ nên áp dụng trong chương trình có ít lần sắp xếp và với kích thước n nhỏ. Về tốc độ, BubbleSort luôn đứng bét, nhưng mã lệnh của nó lại hết sức đơn giản mà người mới học lập trình nào cũng có thể cài đặt được, tính ổn định của BubbleSort cũng rất đáng chú ý. Trong những thuật toán có độ phức tạp $O(n^2)$, InsertionSort tỏ ra nhanh hơn những phương pháp còn lại và cũng có tính ổn định, mã lệnh cũng tương đối đơn giản, dễ nhớ. SelectionSort thì không ổn định nhưng với n nhỏ, việc chọn ra m khoá nhỏ nhất có thể thực hiện dễ dàng chứ không cần phải sắp xếp lại toàn bộ như sắp xếp chèn.

Thuật toán đếm phân phối và thuật toán sắp xếp bằng cơ số nên được tận dụng trong trường hợp các khoá sắp xếp là số tự nhiên (hay là một kiểu dữ liệu có thể quy ra thành các số tự nhiên) bởi những thuật toán này có tốc độ rất cao. Thuật toán sắp xếp bằng cơ số cũng có thể sắp xếp dãy khoá có số thực hay số âm nhưng cần đưa vào một số sửa đổi nhỏ.

QuickSort, HeapSort, MergeSort và ShellSort là những thuật toán sắp xếp tổng quát, dãy khoá thuộc kiểu dữ liệu có thứ tự nào cũng có thể áp dụng được chứ không nhất thiết phải là các số.

QuickSort gặp nhược điểm trong trường hợp suy biến nhưng xác suất xảy ra trường hợp này rất nhỏ. HeapSort thì mã lệnh hơi phức tạp và khó nhớ, nhưng nếu cần chọn ra m khoá lớn nhất trong dãy khoá thì dùng HeapSort sẽ không phải sắp xếp lại toàn bộ dãy. MergeSort phải đòi hỏi thêm một không gian nhớ phụ, nên áp dụng nó trong trường hợp sắp xếp trên file. Còn ShellSort thì hơi khó trong việc đánh giá về thời gian thực thi, nó là sửa đổi của thuật toán sắp xếp chèn nhưng lại có tốc độ tương đối tốt, mã lệnh đơn giản và lượng bộ nhớ cần huy động rất ít. Tuy nhiên, những nhược điểm của bốn phương pháp này quá nhỏ so với ưu điểm chung của chúng là nhanh. Hơn nữa, chúng được đánh giá cao không chỉ vì tính tổng quát và tốc độ nhanh, mà còn là kết quả của những cách tiếp cận khoa học đối với bài toán sắp xếp.

Những thuật toán trên không chỉ đơn thuần là cho ta hiểu thêm về một cách sắp xếp mới, mà việc cài đặt chúng cũng cho chúng ta thêm nhiều kinh nghiệm: Kỹ thuật sử dụng số ngẫu nhiên, kỹ thuật "chia để trị", kỹ thuật dùng các biến với vai trò luân phiên v.v... Vậy nên nắm vững nội dung của những thuật toán đó, mà cách thuộc tốt nhất chính là cài đặt chúng vài lần với các ràng buộc dữ liệu khác nhau (nếu có thể thử được trên hai ngôn ngữ lập trình thì rất tốt) và cũng đừng quên kỹ thuật sắp xếp bằng chỉ số.

Bài tập

Bài 1

Tìm hiểu các tài liệu khác để chứng minh rằng: Bất cứ thuật toán sắp xếp tổng quát nào dựa trên phép so sánh giá trị hai khoá đều có độ phức tạp tính toán trong trường hợp xấu nhất là $\Omega(n \lg n)$. (Sử dụng mô hình cây quyết định – Decision Tree Model).

Bài 2

Viết thuật toán QuickSort không đệ quy

Bài 3

Cho một danh sách thí sinh gồm n người, mỗi người cho biết tên và điểm thi, hãy chọn ra m người điểm cao nhất. Giải quyết bằng thuật toán có độ phức tạp O(n).

Bài 4

Có 2 tính chất quan trọng của thuật toán sắp xếp: Stability và In place:

- ❖ Stability: Tính ổn định
- ❖ In place: Giải thuật không yêu cầu thêm không gian nhớ phụ, điều này cho phép sắp xếp một số lượng lớn các khoá mà không cần cấp phát thêm bộ nhớ. (Tuy vậy việc cấp phát thêm một lượng bộ nhớ cỡ $\Theta(1)$ vẫn được cho phép)

Trong những thuật toán ta đã khảo sát, những thuật toán nào là stability, thuật toán nào là in place, thuật toán nào có cả hai tính chất trên.

Bài 5

Cho một mảng a[1..n]

- Tìm giá trị xuất hiện nhiều hơn $n/2$ lần trong a hoặc thông báo rằng không tồn tại giá trị như vậy. Tìm giải thuật với độ phức tạp O(n)

b) Cho số tự nhiên k , Liệt kê tất cả các giá trị xuất hiện nhiều hơn n/k lần trong a . Tìm giải thuật với độ phức tạp $O(n \cdot k)$

Cách giải:

a) Nếu một giá trị xuất hiện nhiều hơn $n/2$ lần trong a , giá trị đó phải là trung vị của dãy a . Ta sẽ tìm trung vị của dãy và duyệt lại dãy một lần nữa để xác nhận trung vị có xuất hiện nhiều hơn $n/2$ không.

b) Nếu một giá trị xuất hiện nhiều hơn n/k lần trong a , thì khi sắp xếp dãy a theo thứ tự không giảm, giá trị đó phải nằm ở một trong các vị trí $\frac{i^* n}{k}$ ($1 \leq i \leq k$). Áp dụng thuật toán thứ tự thông kê để tìm phần tử lớn thứ $n/k, 2n/k, \dots, (k-1)n/k, n$, và duyệt lại dãy để xác định giá trị của phần tử nào xuất hiện nhiều hơn n/k lần.

Bài 6

Thuật toán sắp xếp bằng cơ sở trực tiếp có ổn định không ? Tại sao ?

Bài 7

Cài đặt thuật toán sắp xếp trộn hai đường tự nhiên

Bài 8

Tìm hiểu phép trộn k đường và các phương pháp sắp xếp ngoài (trên tệp truy nhập tuần tự và tệp truy nhập ngẫu nhiên)

§9. TÌM KIẾM (SEARCHING)

9.1. BÀI TOÁN TÌM KIẾM

Cùng với sắp xếp, tìm kiếm là một đòi hỏi rất thường xuyên trong các ứng dụng tin học. Bài toán tìm kiếm có thể phát biểu như sau:

Cho một dãy gồm n bản ghi $r[1..n]$. Mỗi bản ghi $r[i]$ ($1 \leq i \leq n$) tương ứng với một khoá $k[i]$. Hãy tìm bản ghi có giá trị khoá bằng X cho trước.

X được gọi là khoá tìm kiếm hay đối trị tìm kiếm (argument).

Công việc tìm kiếm sẽ hoàn thành nếu như có một trong hai tình huống sau xảy ra:

- ❖ Tìm được bản ghi có khoá tương ứng bằng X, lúc đó phép tìm kiếm thành công.
- ❖ Không tìm được bản ghi nào có khoá tìm kiếm bằng X cả, phép tìm kiếm thất bại.

Tương tự như sắp xếp, ta coi khoá của một bản ghi là đại diện cho bản ghi đó. Và trong một số thuật toán sẽ trình bày dưới đây, ta coi kiểu dữ liệu cho mỗi khoá cũng có tên gọi là TKey.

```
const
  n = ...; {Số khoá trong dãy khoá, có thể khai dưới dạng biến số nguyên để tùy biến hơn}
type
  TKey = ...; {Kiểu dữ liệu một khoá}
  TArray = array[1..n] of TKey;
var
  k: TArray; {Dãy khoá}
```

9.2. TÌM KIẾM TUẦN TỤ (SEQUENTIAL SEARCH)

Tìm kiếm tuần tự là một kỹ thuật tìm kiếm đơn giản. Nội dung của nó như sau: Bắt đầu từ bản ghi đầu tiên, lần lượt so sánh khoá tìm kiếm với khoá tương ứng của các bản ghi trong danh sách, cho tới khi tìm thấy bản ghi mong muốn hoặc đã duyệt hết danh sách mà chưa thấy

```
{Tìm kiếm tuần tự trên dãy khoá k[1..n]; hàm này thử tìm xem trong dãy có khoá nào = X không, nếu thấy nó trả về chỉ số
của khoá ấy, nếu không thấy nó trả về 0. Có sử dụng một khoá phụ k[n+1] được gán giá trị = X}
function SequentialSearch(X: TKey): Integer;
var
  i: Integer;
begin
  i := 1;
  while (i <= n) and (k[i] ≠ X) do i := i + 1;
  if i = n + 1 then SequentialSearch := 0
  else SequentialSearch := i;
end;
```

Để thấy rằng độ phức tạp của thuật toán tìm kiếm tuần tự trong trường hợp tốt nhất là $O(1)$, trong trường hợp xấu nhất là $O(n)$ và trong trường hợp trung bình là $\Theta(n)$.

9.3. TÌM KIẾM NHỊ PHÂN (BINARY SEARCH)

Phép tìm kiếm nhị phân có thể áp dụng trên dãy khoá đã có thứ tự: $k[1] \leq k[2] \leq \dots \leq k[n]$.

Giả sử ta cần tìm trong đoạn $k[\inf..\sup]$ với khoá tìm kiếm là X, trước hết ta xét khoá nằm giữa dãy $k[\text{median}]$ với $\text{median} = (\inf + \sup) \text{ div } 2$;

- ❖ Nếu $k[\text{median}] < X$ thì có nghĩa là đoạn từ $k[\text{inf}]$ tới $k[\text{median}]$ chỉ chứa toàn khoá $< X$, ta tiếp tục hành tìm kiếm tiếp với đoạn từ $k[\text{median}+1]$ tới $k[\text{sup}]$.
- ❖ Nếu $k[\text{median}] > X$ thì có nghĩa là đoạn từ $k[\text{median}]$ tới $k[\text{sup}]$ chỉ chứa toàn khoá $> X$, ta tiếp tục hành tìm kiếm tiếp với đoạn từ $k[\text{inf}]$ tới $k[\text{median}-1]$.
- ❖ Nếu $k[\text{median}] = X$ thì việc tìm kiếm thành công (kết thúc quá trình tìm kiếm).

Quá trình tìm kiếm sẽ thất bại nếu đến một bước nào đó, đoạn tìm kiếm là rỗng ($\text{inf} > \text{sup}$).

{Tìm kiếm nhị phân trên dãy khoá $k[1] \leq k[2] \leq \dots \leq k[n]$; hàm này thử tìm xem trong dãy có khoá nào = X không, nếu thấy nó trả về chỉ số của khoá ấy, nếu không thấy nó trả về 0}


```
function BinarySearch(X: TKey): Integer;
var
  inf, sup, median: Integer;
begin
  inf := 1; sup := n;
  while inf ≤ sup do
    begin
      median := (inf + sup) div 2;
      if k[median] = X then
        begin
          BinarySearch := median;
          Exit;
        end;
      if k[median] < X then inf := median + 1
      else sup := median - 1;
    end;
  BinarySearch := 0;
end;
```

Người ta đã chứng minh được độ phức tạp tính toán của thuật toán tìm kiếm nhị phân trong trường hợp tốt nhất là $O(1)$, trong trường hợp xấu nhất là $O(\lg n)$ và trong trường hợp trung bình là $O(\lg n)$. Tuy nhiên, ta không nên quên rằng trước khi sử dụng tìm kiếm nhị phân, dãy khoá phải được sắp xếp rồi, tức là thời gian chi phí cho việc sắp xếp cũng phải tính đến. Nếu dãy khoá luôn luôn biến động bởi phép bổ sung hay loại bỏ đi thì lúc đó chi phí cho sắp xếp lại nổi lên rất rõ làm bộc lộ nhược điểm của phương pháp này.

9.4. CÂY NHỊ PHÂN TÌM KIẾM (BINARY SEARCH TREE - BST)

Cho n khoá $k[1..n]$, trên các khoá có quan hệ thứ tự toàn phần. Cây nhị phân tìm kiếm ứng với dãy khoá đó là một cây nhị phân mà mỗi nút chứa giá trị một khoá trong n khoá đã cho, hai giá trị chứa trong hai nút bất kỳ là khác nhau. Đối với mọi nút trên cây, tính chất sau luôn được thoả mãn:

- ❖ Mọi khoá nằm trong cây con trái của nút đó đều nhỏ hơn khoá ứng với nút đó.
- ❖ Mọi khoá nằm trong cây con phải của nút đó đều lớn hơn khoá ứng với nút đó

**Hình 38: Cây nhị phân tìm kiếm**

Thuật toán tìm kiếm trên cây có thể mô tả chung như sau:

Trước hết, khoá tìm kiếm X được so sánh với khoá ở gốc cây, và 4 tình huống có thể xảy ra:

- ❖ Không có gốc (cây rỗng): X không có trên cây, phép tìm kiếm thất bại
- ❖ X trùng với khoá ở gốc: Phép tìm kiếm thành công
- ❖ X nhỏ hơn khoá ở gốc, phép tìm kiếm được tiếp tục trong cây con trái của gốc với cách làm tương tự
- ❖ X lớn hơn khoá ở gốc, phép tìm kiếm được tiếp tục trong cây con phải của gốc với cách làm tương tự

Giả sử cấu trúc một nút của cây được mô tả như sau:

```

type
  PNode = ^TNode; {Con trỏ chua liên kết tới một nút}
  TNode = record {Cấu trúc nút}
    Info: TKey; {Trường chua khoá}
    Left, Right: PNode; {con trỏ tới nút con trái và phải, trỏ tới nil nếu không có nút con trái (phải)}
  end;
  Gốc của cây được lưu trong con trỏ Root. Cây rỗng thì Root = nil
  
```

Thuật toán tìm kiếm trên cây nhị phân tìm kiếm có thể viết như sau:

{Hàm tìm kiếm trên BST, nó trả về nút chua khoá tìm kiếm X nếu tìm thấy, trả về nil nếu không tìm thấy}

```

function BSTSearch(X: TKey): PNode;
var
  p: PNode;
begin
  p := Root; {Bắt đầu với nút gốc}
  while p ≠ nil do
    if X = p^.Info then Break;
    else
      if X < p^.Info then p := p^.Left
      else p := p^.Right;
  BSTSearch := p;
end;
  
```


Thuật toán dựng cây nhị phân tìm kiếm từ dãy khoá k[1..n] cũng được làm gần giống quá trình tìm kiếm. Ta chèn lần lượt các khoá vào cây, trước khi chèn, ta tìm xem khoá đó đã có trong cây hay chưa, nếu đã có rồi thì bỏ qua, nếu nó chưa có thì ta thêm nút mới chua khoá cần chèn và nối nút đó vào cây nhị phân tìm kiếm tại mồi liên kết vừa rẽ sang khiến quá trình tìm kiếm thất bại.

```
{Thủ tục chèn khoá X vào BST}
procedure BSTInsert(X);
var
  p, q: PNode;
begin
  q := nil; p := Root; {Bắt đầu với p = nút gốc; q là con trỏ chạy đuôi theo sau}
  while p ≠ nil do
    begin
      q := p;
      if X = p^.Info then Break;
      else {X ≠ p^.Info thì cho p chạy sang nút con, q^ luôn giữ vai trò là cha của p^}
        if X < p^.Info then p := p^.Left
        else p := p^.Right;
    end;
  if p = nil then {Khoá X chưa có trong BST}
    begin
      New(p); {Tạo nút mới}
      p^.Info := X; {Đưa giá trị X vào nút mới tạo ra}
      p^.Left := nil; p^.Right := nil; {Nút mới khi chèn vào BST sẽ trở thành nút lá}
      if Root = nil then Root := NewNode {BST đang rỗng, đặt Root là nút mới tạo}
      else {Móc NewNode^ vào nút cha q^}
        if X < q^.Info then q^.Left := NewNode
        else q^.Right := NewNode;
    end;
  end;
```

Phép loại bỏ trên cây nhị phân tìm kiếm không đơn giản như phép bổ sung hay phép tìm kiếm.


Muốn xoá một giá trị trong cây nhị phân tìm kiếm (Tức là dựng lại cây mới chứa tất cả những giá trị còn lại), trước hết ta tìm xem giá trị cần xoá nằm ở nút D nào, có ba khả năng xảy ra:

- ❖ Nút D là nút lá, trường hợp này ta chỉ việc đem mỗi nối cũ trỏ tới nút D (từ nút cha của D) thay bởi nil, và giải phóng bộ nhớ cấp cho nút D (Hình 39).



Hình 39: Xóa nút lá ở cây BST


- ❖ Nút D chỉ có một nhánh con, khi đó ta đem nút gốc của nhánh con đó thê vào chỗ nút D, tức là chỉnh lại mối nối: Từ nút cha của nút D không nối tới nút D nữa mà nối tới nhánh con duy nhất của nút D. Cuối cùng, ta giải phóng bộ nhớ đã cấp cho nút D (Hình 40)



Hình 40. Xóa nút chỉ có một nhánh con trên cây BST


❖ Nút D có cả hai nhánh con trái và phải, khi đó có hai cách làm đều hợp lý cả:

Hoặc tìm nút chứa khoá lớn nhất trong cây con trái, đưa giá trị chứa trong đó sang nút D, rồi xoá nút này. Do tính chất của cây BST, nút chứa khoá lớn nhất trong cây con trái chính là nút cực phải của cây con trái nên nó không thể có hai con được, việc xoá đưa về hai trường hợp trên (Hình 41)



Hình 41: Xóa nút có cả hai nhánh con trên cây BST thay bằng nút cực phải của cây con trái

Hoặc tìm nút chứa khoá nhỏ nhất trong cây con phải, đưa giá trị chứa trong đó sang nút D, rồi xoá nút này. Do tính chất của cây BST, nút chứa khoá nhỏ nhất trong cây con phải chính là nút cực trái của cây con phải nên nó không thể có hai con được, việc xoá đưa về hai trường hợp trên.



Hình 42: Xóa nút có cả hai nhánh con trên cây BST thay bằng nút cực trái của cây con phải

```

{Thủ tục xoá khoá X khỏi BST}
procedure BSTDelete(X: TKey);
var
  p, q, Node, Child: PNode;
begin
  p := Root; q := nil; {Về sau, khi p trở sang nút khác, ta luôn giữ cho q^ luôn là cha của p^}
  while p ≠ nil do {Tìm xem trong cây có khoá X không?}
    begin
      if p^.Info = X then Break; {Tim thấy}
      q := p;
      if X < p^.Info then p := p^.Left
      else p := p^.Right;
    end;
  if p = nil then Exit; {X không tồn tại trong BST nên không xoá được}
  if (p^.Left ≠ nil) and (p^.Right ≠ nil) then {p^ có cả con trái và con phải}
    begin
      Node := p; {Giữ lại nút chứa khoá X}
      q := p; p := p^.Left; {Chuyển sang nhánh con trái để tìm nút cực phải}
      while p^.Right ≠ nil do
        begin
          q := p; p := p^.Right;
        end;
      Node^.Info := p^.Info; {Chuyển giá trị từ nút cực phải trong nhánh con trái lên Node^}
    end;
  {Nút bị xoá giờ đây là nút p^, nó chỉ có nhiều nhất một con}
  {Nếu p^ có một nút con thì đem Child trả tới nút con đó, nếu không có thì Child = nil}
  if p^.Left ≠ nil then Child := p^.Left
  else Child := p^.Right;
  if p = Root then Root := Child; {Nút p^ bị xoá là gốc cây}
  else {Nút bị xoá p^ không phải gốc cây thì lấy mỗi nối từ cha của nó là q^ nối thẳng tới Child}
    if q^.Left = p then q^.Left := Child
    else q^.Right := Child;
  Dispose(p);
end;

```

Trường hợp trung bình, thì các thao tác tìm kiếm, chèn, xoá trên BST có độ phức tạp là $O(lgn)$. Còn trong trường hợp xấu nhất, cây nhị phân tìm kiếm bị suy biến thì các thao tác đó đều có độ phức tạp là $O(n)$, với n là số nút trên cây BST.


Nếu ta mở rộng hơn khái niệm cây nhị phân tìm kiếm như sau: Giá trị lưu trong một nút lớn hơn **hoặc bằng** các giá trị lưu trong cây con trái và nhỏ hơn các giá trị lưu trong cây con phải. Thì chỉ cần sửa đổi thủ tục BSTInsert một chút, khi chèn lần lượt vào cây n giá trị, cây BST sẽ có n nút (có thể có hai nút chứa cùng một giá trị). Khi đó nếu ta duyệt các nút của cây theo kiểu trung thứ tự (inorder traversal), ta sẽ liệt kê được các giá trị lưu trong cây theo thứ tự tăng dần. Phương pháp sắp xếp này người ta gọi là Tree Sort. Độ phức tạp tính toán trung bình của Tree Sort là $O(nlgn)$.

Phép tìm kiếm trên cây BST sẽ kém hiệu quả nếu như cây bị suy biến, người ta có nhiều cách xoay xở để tránh trường hợp này. Đó là phép quay cây để dựng cây nhị phân cân đối AVL, hay kỹ thuật dựng cây nhị phân tìm kiếm tối ưu. Những kỹ thuật này ta có thể tham khảo trong các tài liệu khác về cấu trúc dữ liệu và giải thuật.

9.5. PHÉP BĂM (HASH)

Tư tưởng của phép băm là dựa vào giá trị các khoá $k[1..n]$, chia các khoá đó ra thành các nhóm. **Những khoá thuộc cùng một nhóm có một đặc điểm chung** và đặc điểm này không có trong các nhóm khác. Khi có một khoá tìm kiếm X, trước hết ta xác định xem nếu X thuộc vào dãy khoá đã cho thì nó phải thuộc nhóm nào và tiến hành tìm kiếm trên nhóm đó.

Một ví dụ là trong cuốn từ điển, các bạn sinh viên thường dán vào 26 mảnh giấy nhỏ vào các trang để đánh dấu trang nào là trang khởi đầu của một đoạn chứa các từ có cùng chữ cái đầu. Để khi tra từ chỉ cần tìm trong các trang chứa những từ có cùng chữ cái đầu với từ cần tìm.



Một ví dụ khác là trên dãy các khoá số tự nhiên, ta có thể chia nó là làm m nhóm, mỗi nhóm gồm các khoá đồng dư theo mô-đun m.

Có nhiều cách cài đặt phép băm:

- ❖ Cách thứ nhất là chia dãy khoá làm các đoạn, mỗi đoạn chứa những khoá thuộc cùng một nhóm và ghi nhận lại vị trí các đoạn đó. Để khi có khoá tìm kiếm, có thể xác định được ngay cần phải tìm khoá đó trong đoạn nào.
- ❖ Cách thứ hai là chia dãy khoá làm m nhóm, Mỗi nhóm là một danh sách nối đơn chứa các giá trị khoá và ghi nhận lại chốt của mỗi danh sách nối đơn. Với một khoá tìm kiếm, ta xác định được phải tìm khoá đó trong danh sách nối đơn nào và tiến hành tìm kiếm tuần tự trên danh sách nối đơn đó. Với cách lưu trữ này, việc bổ sung cũng như loại bỏ một giá trị khỏi tập hợp khoá dễ dàng hơn rất nhiều phương pháp trên.
- ❖ Cách thứ ba là nếu chia dãy khoá làm m nhóm, mỗi nhóm được lưu trữ dưới dạng cây nhị phân tìm kiếm và ghi nhận lại gốc của các cây nhị phân tìm kiếm đó, phương pháp này có thể nói là tốt hơn hai phương pháp trên, tuy nhiên dãy khoá phải có quan hệ thứ tự toàn phần thì mới làm được.

9.6. KHOÁ SỐ VỚI BÀI TOÁN TÌM KIẾM

Mọi dữ liệu lưu trữ trong máy tính đều được số hoá, tức là đều được lưu trữ bằng các đơn vị Bit, Byte, Word v.v... Điều đó có nghĩa là một giá trị khoá bất kỳ, ta hoàn toàn có thể biết được nó được mã hoá bằng con số như thế nào. Và một điều chắc chắn là hai khoá khác nhau sẽ được lưu trữ bằng hai số khác nhau.

Đối với bài toán sắp xếp, ta không thể đưa việc sắp xếp một dãy khoá bất kỳ về việc sắp xếp trên một dãy khoá số là mã của các khoá. Bởi quan hệ thứ tự trên các con số đó có thể khác với thứ tự cần sắp của các khoá.


Nhưng đối với bài toán tìm kiếm thì khác, với một khoá tìm kiếm, câu trả lời hoặc là “Không tìm thấy” hoặc là “Có tìm thấy và ở chỗ ...” nên ta hoàn toàn có thể thay các khoá bằng các mã số của nó mà không bị sai lầm, chỉ lưu ý một điều là: hai khoá khác nhau phải mã hoá thành hai số khác nhau mà thôi.

Nói như vậy có nghĩa là việc nghiên cứu những thuật toán tìm kiếm trên các dãy khoá số rất quan trọng, và dưới đây ta sẽ trình bày một số phương pháp đó.

9.7. CÂY TÌM KIẾM SỐ HỌC (DIGITAL SEARCH TREE - DST)

Xét dãy khoá $k[1..n]$ là các số tự nhiên, mỗi giá trị khoá khi đổi ra hệ nhị phân có z chữ số nhị phân (bit), các bit này được đánh số từ 0 (là hàng đơn vị) tới $z - 1$ từ phải sang trái.

Ví dụ:




Hình 43: Đánh số các bit

Cây tìm kiếm số học chứa các giá trị khoá này có thể mô tả như sau: Trước hết, nó là một cây nhị phân mà mỗi nút chứa một giá trị khoá. Nút gốc có tối đa hai cây con, ngoài giá trị khoá chứa ở nút gốc, tất cả những giá trị khoá có bit cao nhất là 0 nằm trong cây con trái, còn tất cả những giá trị khoá có bit cao nhất là 1 nằm ở cây con phải. Đối với hai nút con của nút gốc, vẫn đề tương tự đối với bit $z - 2$ (bit đứng thứ nhì từ trái sang).

So sánh cây tìm kiếm số học với cây nhị phân tìm kiếm, chúng chỉ khác nhau về cách chia hai cây con trái/phải. Đối với cây nhị phân tìm kiếm, việc chia này được thực hiện bằng cách so sánh với khoá nằm ở nút gốc, còn đối với cây tìm kiếm số học, nếu nút gốc có mức là d thì việc chia cây con được thực hiện theo bit thứ d tính từ trái sang (bit $z - d$) của mỗi khoá.

Ta nhận thấy rằng những khoá bắt đầu bằng bit 0 chắc chắn nhỏ hơn những khoá bắt đầu bằng bit 1, đó là điểm tương đồng giữa cây nhị phân tìm kiếm và cây tìm kiếm số học: Với mỗi nút nhánh: Mọi giá trị chứa trong cây con trái đều nhỏ hơn giá trị chứa trong cây con phải (Hình 44).



Hình 44: Cây tìm kiếm số học

Giả sử cấu trúc một nút của cây được mô tả như sau:

```
type
  PNode = ^TNode; {Con trỏ chua liên kết tới một nút}
  TNode = record {Cấu trúc nút}
    Info: TKey; {Trường chua khoá}
    Left, Right: PNode; {con trỏ tới nút con trái và phải, trỏ tới nil nếu không có nút con trái (phải)}
  end;
Gốc của cây được lưu trong con trỏ Root. Ban đầu nút Root = nil (cây rỗng)
```

Với khoá tìm kiếm X, việc tìm kiếm trên cây tìm kiếm số học có thể mô tả như sau: Ban đầu đứng ở nút gốc, xét lần lượt các bit của X từ trái sang phải (từ bit z - 1 tới bit 0), nếu gặp bit bằng 0 thì rẽ sang nút con trái, nếu gặp bit bằng 1 thì rẽ sang nút con phải. Quá trình cứ tiếp tục như vậy cho tới khi gặp một trong hai tình huống sau:

- ❖ Đi tới một nút rỗng (do rẽ theo một liên kết nil), quá trình tìm kiếm thất bại do khoá X không có trong cây.
- ❖ Đi tới một nút mang giá trị đúng bằng X, quá trình tìm kiếm thành công

```
{Hàm tìm kiếm trên cây tìm kiếm số học, nó trả về nút chua khoá tìm kiếm X nếu tìm thấy, trả về nil nếu không tìm thấy. z là độ dài dãy bit biểu diễn một khoá}
function DSTSearch(X: TKey): PNode;
var
  b: Integer;
  p: PNode;
begin
  b := z; p := Root; {Bắt đầu với nút gốc}
  while (p ≠ nil) and (p^.Info ≠ X) do {Chưa gặp phải một trong 2 tình huống trên}
    begin
      b := b - 1; {Xét bit b của X}
      if <Bit b của X là 0> then p := p^.Left {Gặp 0 rẽ trái}
      else p := p^.Right; {Gặp 1 rẽ phải}
    end;
  DSTSearch := p;
end;
```

Thuật toán dựng cây tìm kiếm số học từ dãy khoá k[1..n] cũng được làm gần giống quá trình tìm kiếm. Ta chèn lần lượt các khoá vào cây, trước khi chèn, ta tìm xem khoá đó đã có trong cây hay chưa, nếu đã có rồi thì bỏ qua, nếu nó chưa có thì ta thêm nút mới chua khoá cần chèn và nối nút đó vào cây tìm kiếm số học tại mỗi nối rỗng vừa rẽ sang khiến quá trình tìm kiếm thất bại

```
{Thủ tục chèn khoá X vào cây tìm kiếm số học}
procedure DSTInsert(X: TKey);
var
  b: Integer;
  p, q: PNode;
begin
  begin
    b := z;
    p := Root;
    while (p ≠ nil) and (p^.Info ≠ X) do
      begin
        b := b - 1; {Xét bit b của X}
        q := p; {Khi p chạy xuống nút con thì q^ luôn giữ vai trò là nút cha của p^}
        if <Bit b của X là 0> then p := p^.Left {Gấp 0 rẽ trái}
        else p := p^.Right; {Gấp 1 rẽ phải}
      end;
    if p = nil then {Giá trị X chưa có trong cây}
      begin
        New(p); {Tạo ra một nút mới p^}
        p^.Info := X; {Nút mới tạo ra sẽ chứa khoá X}
        p^.Left := nil; p^.Right := nil; {Nút mới đó sẽ trở thành một lá của cây}
        if Root = nil then Root := p {Cây đang là rỗng thì nút mới thêm trở thành gốc}
        else {Không thì mốc p^ vào mỗi nỗi vừa rẽ sang từ q^}
          if <Bit b của X là 0> then q^.Left := p
          else q^.Right := p;
      end;
  end;
end;
```

Muốn xoá bỏ một giá trị khỏi cây tìm kiếm số học, trước hết ta xác định nút chứa giá trị cần xoá là nút D nào, sau đó tìm trong nhánh cây gốc D ra một nút lá bất kỳ, chuyển giá trị chứa trong nút lá đó sang nút D rồi xoá nút lá.

```
{Thủ tục xoá khoá X khỏi cây tìm kiếm số học}
procedure DSTDelete(X: TKey);
var
  b: Integer;
  p, q, Node: PNode;
begin
  {Trước hết, tìm kiếm giá trị X xem nó nằm ở nút nào}
  b := z;
  p := Root;
  while (p ≠ nil) and (p^.Info ≠ X) do
    begin
      b := b - 1;
      q := p; {Mỗi lần p chuyển sang nút con, ta luôn đảm bảo cho q^ là nút cha của p^}
      if <Bit b của X là 0> then p := p^.Left
      else p := p^.Right;
    end;
  if p = nil then Exit; {X không tồn tại trong cây thì không xoá được}
  Node := p; {Giữ lại nút chứa khoá cần xoá}
  while (p^.Left ≠ nil) or (p^.Right ≠ nil) do {chừng nào p^ chưa phải là lá}
    begin
      q := p; {q chạy đuổi theo p, còn p chuyển xuống một trong 2 nhánh con}
      if p^.Left ≠ nil then p := p^.Left
      else p := p^.Right;
    end;
  Node^.Info := p^.Info; {Chuyển giá trị từ nút lá p^ sang nút Node^}
  if Root = p then Root := nil {Cây chỉ gồm một nút gốc và bây giờ xoá cả gốc}
  else {Cắt mỗi nỗi từ q^ tới p^}
    if q^.Left = p then q^.Left := nil
    else q^.Right := nil;
  Dispose(p);
end;
```

Về mặt trung bình, các thao tác tìm kiếm, chèn, xoá trên cây tìm kiếm số học đều có độ phức tạp là $O(\min(z, \lg n))$ còn trong trường hợp xấu nhất, độ phức tạp của các thao tác đó là $O(z)$, bởi cây tìm kiếm số học có chiều cao không quá $z + 1$.

9.8. CÂY TÌM KIẾM CƠ SỐ (RADIX SEARCH TREE - RST)

Trong cây tìm kiếm số học, cũng như cây nhị phân tìm kiếm, phép tìm kiếm tại mỗi bước phải so sánh giá trị khoá X với giá trị lưu trong một nút của cây. Trong trường hợp các khoá có cấu trúc lớn, việc so sánh này có thể mất nhiều thời gian.


Cây tìm kiếm cơ số là một phương pháp khắc phục nhược điểm đó, nội dung của nó có thể tóm tắt như sau:

Trong cây tìm kiếm cơ số là một cây nhị phân, chỉ có nút lá chứa giá trị khoá, còn giá trị chưa trong các nút nhánh là vô nghĩa. Các nút lá của cây tìm kiếm cơ số đều nằm ở mức $z + 1$.

Đối với nút gốc của cây tìm kiếm cơ số, nó có tối đa hai nhánh con, mọi khoá chứa trong nút lá của nhánh con trái đều có bit cao nhất là 0, mọi khoá chứa trong nút lá của nhánh con phải đều có bit cao nhất là 1.

Đối với hai nhánh con của nút gốc, vẫn đề tương tự với bit thứ $z - 2$, ví dụ với nhánh con trái của nút gốc, nó lại có tối đa hai nhánh con, mọi khoá chứa trong nút lá của nhánh con trái đều có bit thứ $z - 2$ là 0 (chúng bắt đầu bằng hai bit 00), mọi khoá chứa trong nút lá của nhánh con phải đều có bit thứ $z - 2$ là 1 (chúng bắt đầu bằng hai bit 01)...

Tổng quát với nút ở mức d , nó có tối đa hai nhánh con, mọi nút lá của nhánh con trái chứa khoá có bit $z - d$ là 0, mọi nút lá của nhánh con phải chứa khoá có bit $z - d$ là 1 (Hình 45).



Hình 45: Cây tìm kiếm cơ số

Khác với cây nhị phân tìm kiếm hay cây tìm kiếm số học. Cây tìm kiếm cơ số được khởi tạo gồm có một nút gốc, và **nút gốc tồn tại trong suốt quá trình sử dụng**: nó không bao giờ bị xoá đi cả.


Để tìm kiếm một giá trị X trong cây tìm kiếm cơ sở, ban đầu ta đứng ở nút gốc và duyệt dãy bit của X từ trái qua phải (từ bit $z - 1$ đến bit 0), gặp bit bằng 0 thì rẽ sang nút con trái còn gặp bit bằng 1 thì rẽ sang nút con phải, cứ tiếp tục như vậy cho tới khi một trong hai tình huống sau xảy ra:

- ❖ Hoặc đi tới một nút rỗng (do rẽ theo liên kết nil) quá trình tìm kiếm thất bại do X không có trong RST
- ❖ Hoặc đã duyệt hết dãy bit của X và đang đứng ở một nút lá, quá trình tìm kiếm thành công vì chắc chắn nút lá đó chứa giá trị đúng bằng X.

```
{Hàm tìm kiếm trên cây tìm kiếm cơ sở, trả về nút lá chứa khóa tìm kiếm X nếu tìm thấy, trả về nil nếu không tìm thấy. z là độ dài dãy bit biểu diễn một khoá}
function RSTSearch(X: TKey): PNode;
var
  b: Integer;
  p: PNode;
begin
  b := z; p := Root; {Bắt đầu với nút gốc, đối với RST thì gốc luôn có sẵn}
  repeat
    b := b - 1; {Xét bit b của X}
    if <Bit b của X là 0> then p := p^.Left {Gặp 0 rẽ trái}
    else p := p^.Right; {Gặp 1 rẽ phải}
  until (p = nil) or (b = 0);
  RSTSearch := p;
end;
```

Thao tác chèn một giá trị X vào RST được thực hiện như sau: Đầu tiên, ta đứng ở gốc và duyệt dãy bit của X từ trái qua phải (từ bit $z - 1$ về bit 0), cứ gặp 0 thì rẽ trái, gặp 1 thì rẽ phải. Nếu quá trình rẽ theo một liên kết nil (đi tới nút rỗng) thì lập tức tạo ra một nút mới, và nối vào theo liên kết đó để có đường đi tiếp. Sau khi duyệt hết dãy bit của X, ta sẽ dừng lại ở một nút lá của RST, và công việc cuối cùng là đặt giá trị X vào nút lá đó.


Ví dụ:



Hình 46: Với độ dài dãy bit $z = 3$, cây tìm kiếm cơ sở gồm các khoá 2, 4, 5 và sau khi thêm giá trị 7

```
{Thủ tục chèn khoá X vào cây tìm kiếm cơ sở}
procedure RSTInsert(X: TKey);
var
  b: Integer;
  p, q: PNode;
begin
  b := z; p := Root; {Bắt đầu từ nút gốc, đối với RST thì gốc luôn ≠ nil}
  repeat
    b := b - 1; {Xét bit b của X}
    q := p; {Khi p chạy xuống nút con thì q^ luôn giữ vai trò là nút cha của p^}
    if <Bit b của X là 0> then p := p^.Left {Gặp 0 rẽ trái}
    else p := p^.Right; {Gặp 1 rẽ phải}
    if p = nil then {Không đi được thì đặt thêm nút để đi tiếp}
      begin
        New(p); {Tạo ra một nút mới và đem p trỏ tới nút đó}
        p^.Left := nil; p^.Right := nil;
        if <Bit b của X là 0> then q^.Left := p {Nối p^ vào bên trái q^}
        else q^.Right := p; {Nối p^ vào bên phải q^}
      end;
    until b = 0;
    p^.Info := X; {p^ là nút lá để đặt X vào}
  end;
```

Với cây tìm kiếm cơ sở, việc xoá một giá trị khoá không phải chỉ là xoá riêng một nút lá mà còn phải xoá toàn bộ nhánh độc đạo đi tới nút đó để tránh lãng phí bộ nhớ (Hình 47).



Hình 47: RST chứa các khoá 2, 4, 5, 7 và RST sau khi loại bỏ giá trị 7

Ta lặp lại quá trình tìm kiếm giá trị khoá X, quá trình này sẽ đi từ gốc xuống lá, tại mỗi bước đi, mỗi khi gặp một nút ngã ba (nút có cả con trái và con phải - nút cấp hai), ta ghi nhận lại ngã ba đó và hướng rẽ. Kết thúc quá trình tìm kiếm ta giữ lại được ngã ba đi qua cuối cùng, từ nút đó tới nút lá chứa X là con đường độc đạo (không có chẽ rẽ), ta tiến hành dỡ bỏ tất cả các nút trên đoạn đường độc đạo khỏi cây tìm kiếm cơ sở. Để không bị gặp lỗi khi cây suy biến (không có nút cấp 2) ta coi gốc cũng là nút ngã ba

```

{Thù tục xoá khoá X khỏi cây tìm kiếm cơ sở}
procedure RSTDelete(X: TKey);
var
  b: Integer;
  p, q, TurnNode, Child: PNode;
begin
  {Trước hết, tìm kiếm giá trị X xem nó nằm ở nút nào}
  b := z; p := Root;
  repeat
    b := b - 1;
    q := p; {Mỗi lần p chuyển sang nút con, ta luôn đảm bảo cho q^ là nút cha của p^}
    if <Bit b của X là 0> then p := p^.Left
    else p := p^.Right;
    if (b = z - 1) or (q^.Left ≠ nil) and (q^.Right ≠ nil) then {q^ là nút ngã ba}
      begin
        TurnNode := q; Child := p; {Ghi nhận lại q^ và hướng rẽ}
        end;
    until (p = nil) or (b = 0);
  if p = nil then Exit; {X không tồn tại trong cây thì không xoá được}
  {Trước hết, cắt nhánh độc đạo ra khỏi cây}
  if TurnNode^.Left = Child then TurnNode^.Left := nil
  else TurnNode^.Right := nil;
  p := Child; {Chuyển sang đoạn đường độc đạo, bắt đầu xoá}
  repeat
    q := p;
    {Lưu ý rằng p^ chỉ có tối đa một nhánh con mà thôi, cho p trở sang nhánh con duy nhất nếu có}
    if p^.Left ≠ nil then p := p^.Left
    else p := p^.Right;
    Dispose(q); {Giải phóng bộ nhớ cho nút q^}
  until p = nil;
end;


```

Ta có một nhận xét là: Hình dáng của cây tìm kiếm cơ sở không phụ thuộc vào thứ tự chèn các khoá vào mà chỉ phụ thuộc vào giá trị của các khoá chứa trong cây.


Đối với cây tìm kiếm cơ sở, độ phức tạp tính toán cho các thao tác tìm kiếm, chèn, xoá trong trường hợp xấu nhất cũng như trung bình đều là $O(z)$. Do không phải so sánh giá trị khoá dọc đường đi, nó nhanh hơn cây tìm kiếm số học nếu như gấp các khoá cấu trúc lớn. Tốc độ như vậy có thể nói là tốt, nhưng vấn đề bộ nhớ khiến ta phải xem xét: Giá trị chứa trong các nút nhánh của cây tìm kiếm cơ sở là vô nghĩa dẫn tới sự lãng phí bộ nhớ.

Một giải pháp cho vấn đề này là: Duy trì hai dạng nút trên cây tìm kiếm cơ sở: Dạng nút nhánh chỉ chứa các liên kết trái, phải và dạng nút lá chỉ chứa giá trị khoá. Cài đặt cây này trên một số ngôn ngữ định kiểu quá mạnh đôi khi rất khó.

Giải pháp thứ hai là đặc tả một cây tương tự như RST, nhưng sửa đổi một chút: nếu có nút lá chứa giá trị X được nối với cây bằng một nhánh độc đạo thì cắt bỏ nhánh độc đạo đó, và thay vào chỗ nhánh này chỉ một nút chứa giá trị X. Như vậy các giá trị khoá vẫn chỉ chứa trong các nút lá nhưng các nút lá giờ đây không chỉ nằm trên mức $z + 1$ mà còn nằm trên những mức khác nữa. Phương pháp này không những tiết kiệm bộ nhớ hơn mà còn làm cho quá trình tìm kiếm nhanh hơn. Giá phải trả cho phương pháp này là thao tác chèn, xoá khá phức tạp. Tên của cấu trúc dữ liệu này là Trie (Trie chứ không phải Tree) tìm kiếm cơ sở.



a)



b)

Hình 48: Cây tìm kiếm cơ số a) và Trie tìm kiếm cơ số b)

Tương tự như phương pháp sắp xếp bằng cơ số, phép tìm kiếm bằng cơ số không nhất thiết phải chọn hệ cơ số 2. Ta có thể chọn hệ cơ số lớn hơn để có tốc độ nhanh hơn (kèm theo sự tốn kém bộ nhớ), chỉ lưu ý là cây tìm kiếm số học cũng như cây tìm kiếm cơ số trong trường hợp này không còn là cây nhị phân mà là cây R_phân với R là hệ cơ số được chọn.

Trong các phương pháp tìm kiếm bằng cơ số, thực ra còn một phương pháp tinh tuý và thông minh nhất, nó có cấu trúc gần giống như cây nhưng không có nút dư thừa, và quá trình duyệt bit của khoá tìm kiếm không phải từ trái qua phải mà theo thứ tự của các bit kiểm soát lưu tại mỗi nút đi qua. Phương pháp đó có tên gọi là Practical Algorithm To Retrieve Information Coded In Alphanumeric (PATRICIA) do Morrison đề xuất. Tuy nhiên, việc cài đặt phương pháp này khá phức tạp (đặc biệt là thao tác xoá giá trị khoá), ta có thể tham khảo nội dung của nó trong các tài liệu khác.

9.9. NHỮNG NHẬN XÉT CUỐI CÙNG

Tìm kiếm thường là công việc nhanh hơn sắp xếp nhưng lại được sử dụng nhiều hơn. Trên đây, ta đã trình bày phép tìm kiếm trong một tập hợp để tìm ra bản ghi mang khoá đúng bằng khoá tìm kiếm. Tuy nhiên, người ta có thể yêu cầu tìm bản ghi mang khoá lớn hơn hay nhỏ hơn khoá tìm kiếm, tìm bản ghi mang khoá nhỏ nhất mà lớn hơn khoá tìm kiếm, tìm bản ghi mang khoá lớn nhất mà nhỏ hơn khoá tìm kiếm v.v... Để cài đặt những thuật toán nêu trên cho những trường hợp này cần có một sự mềm dẻo nhất định.

Cũng tương tự như sắp xếp, ta không nên đánh giá giải thuật tìm kiếm này tốt hơn giải thuật tìm kiếm khác. Sử dụng thuật toán tìm kiếm phù hợp với từng yêu cầu cụ thể là kỹ năng của người lập trình, việc cài đặt cây nhị phân tìm kiếm hay cây tìm kiếm cơ sở chỉ để tìm kiếm trên vài chục bản ghi chỉ khẳng định được một điều rõ ràng: không biết thế nào là giải thuật và lập trình.

Bài tập

Bài 1

Hãy thử viết một chương trình SearchDemo tương tự như chương trình SortDemo trong bài trước. Đồng thời viết thêm vào chương trình SortDemo ở bài trước thủ tục TreeSort và đánh giá tốc độ thực của nó.

Bài 2

Tìm hiểu các phương pháp tìm kiếm chuỗi, thuật toán BRUTE-FORCE, thuật toán KNUTH-MORRIS-PRATT, thuật toán BOYER-MOORE và thuật toán RABIN-KARP

Bài 3

Tự tìm hiểu trong các tài liệu khác về tìm kiếm đa hướng (multi-way searching), cây nhị phân AVL, cây (2, 3, 4), cây đũa đen.

Tuy gọi là chuyên đề về “Cấu trúc dữ liệu và giải thuật” nhưng thực ra, ta mới chỉ tìm hiểu về một số cấu trúc dữ liệu và giải thuật hay gặp. Không một tài liệu nào có thể đề cập tới mọi cấu trúc dữ liệu và giải thuật bởi chúng quá phong phú và liên tục được bổ sung. Những cấu trúc dữ liệu và giải thuật không “phổ thông” lắm như lý thuyết đồ thị, hình học, v.v... sẽ được tách ra và sẽ được nói kỹ hơn trong một chuyên đề khác.

Việc đi sâu nghiên cứu những cấu trúc dữ liệu và giải thuật, dù chỉ là một phần nhỏ hẹp cũng nảy sinh rất nhiều vấn đề hay và khó, như các vấn đề lý thuyết về độ phức tạp tính toán, vấn đề NP_đầy đủ v.v... Đó là công việc của những nhà khoa học máy tính. Nhưng trước khi trở thành một nhà khoa học máy tính thì điều kiện cần là phải biết lập trình. Vậy nên khi tìm hiểu bất cứ cấu trúc dữ liệu hay giải thuật nào, nhất thiết ta phải cố gắng cài đặt bằng được. Mọi ý tưởng hay sẽ chỉ là bỏ đi nếu như không biến thành hiệu quả, thực tế là như vậy.



PHẦN 3. QUY HOẠCH ĐỘNG

Các thuật toán đệ quy có ưu điểm dễ cài đặt, tuy nhiên do bản chất của quá trình đệ quy, các chương trình này thường kéo theo những đòi hỏi lớn về không gian bộ nhớ và một khối lượng tính toán khổng lồ.

Quy hoạch động (Dynamic programming) là một kỹ thuật nhằm đơn giản hóa việc tính toán các công thức truy hồi bằng cách lưu trữ toàn bộ hay một phần kết quả tính toán tại mỗi bước với mục đích sử dụng lại. Bản chất của quy hoạch động là thay thế mô hình tính toán “từ trên xuống” (Top-down) bằng mô hình tính toán “từ dưới lên” (Bottom-up).

Từ “programming” ở đây không liên quan gì tới việc lập trình cho máy tính, đó là một thuật ngữ mà các nhà toán học hay dùng để chỉ ra các bước chung trong việc giải quyết một dạng bài toán hay một lớp các vấn đề. Không có một thuật toán tổng quát để giải tất cả các bài toán quy hoạch động.

Mục đích của phần này là cung cấp một cách tiếp cận mới trong việc giải quyết các bài toán tối ưu mang bản chất đệ quy, đồng thời đưa ra các ví dụ để người đọc có thể làm quen và hình thành các kỹ năng trong việc tiếp cận các bài toán quy hoạch động.

§1. CÔNG THỨC TRUY HỒI

1.1. VÍ DỤ

Cho số tự nhiên $n \leq 100$. Hãy cho biết có bao nhiêu cách phân tích số n thành tổng của dãy các số nguyên dương, các cách phân tích là hoán vị của nhau chỉ tính là một cách.

Ví dụ: $n = 5$ có 7 cách phân tích:

1. $5 = 1 + 1 + 1 + 1 + 1$
2. $5 = 1 + 1 + 1 + 2$
3. $5 = 1 + 1 + 3$
4. $5 = 1 + 2 + 2$
5. $5 = 1 + 4$
6. $5 = 2 + 3$
7. $5 = 5$

(Lưu ý: $n = 0$ vẫn coi là có 1 cách phân tích thành tổng các số nguyên dương (0 là tổng của dãy rỗng))

Để giải bài toán này, trong chuyên mục trước ta đã dùng phương pháp liệt kê tất cả các cách phân tích và đếm số câu hình. Nay giờ ta thử nghĩ xem, **có cách nào tính ngay ra số lượng các cách phân tích mà không cần phải liệt kê hay không?** Bởi vì khi số cách phân tích tương đối lớn, phương pháp liệt kê tỏ ra khá chậm. ($n = 100$ có 190569292 cách phân tích).

Nhận xét:

Nếu gọi $F[m, v]$ là số cách phân tích số v thành tổng các số nguyên dương $\leq m$. Khi đó:

Các cách phân tích số v thành tổng các số nguyên dương $\leq m$ có thể chia làm hai loại:

- ❖ Loại 1: Không chứa số m trong phép phân tích, khi đó số cách phân tích loại này chính là số cách phân tích số v thành tổng các số nguyên dương $< m$, tức là số cách phân tích số v thành tổng các số nguyên dương $\leq m - 1$ và bằng $F[m - 1, v]$.
- ❖ Loại 2: Có chứa ít nhất một số m trong phép phân tích. Khi đó nếu trong các cách phân tích loại này ta bỏ đi số m đó thì ta sẽ được các cách phân tích số $v - m$ thành tổng các số nguyên dương $\leq m$ (Lưu ý: điều này chỉ đúng khi không tính lặp lại các hoán vị của một cách). Có nghĩa là về mặt số lượng, số các cách phân tích loại này bằng $F[m, v - m]$

Trong trường hợp $m > v$ thì rõ ràng chỉ có các cách phân tích loại 1, còn trong trường hợp $m \leq v$ thì sẽ có cả các cách phân tích loại 1 và loại 2. Vì thế:

$$F[m, v] = \begin{cases} F[m - 1, v]; & \text{if } m > v \\ F[m - 1, v] + F[m, v - m]; & \text{if } m \leq v \end{cases}$$

Ta có công thức xây dựng $F[m, v]$ từ $F[m - 1, v]$ và $F[m, v - m]$. Công thức này có tên gọi là **công thức truy hồi** đưa việc tính $F[m, v]$ về việc tính các $F[m', v']$ với dữ liệu nhỏ hơn. Tất nhiên cuối cùng ta sẽ quan tâm đến $F[n, n]$: Số các cách phân tích n thành tổng các số nguyên dương $\leq n$.

Ví dụ với $n = 5$, bảng F sẽ là:

F	0	1	2	3	4	5	v
0	1	0	0	0	0	0	
1	1	1	1	1	1	1	
2	1	1	2	2	3	3	
3	1	1	2	3	4	5	
4	1	1	2	3	5	6	
5	1	1	2	3	5	7	
m							

Nhìn vào bảng F, ta thấy rằng $F[m, v]$ được tính bằng tổng của:

Một phần tử ở hàng trên: $F[m - 1, v]$ và một phần tử ở cùng hàng, bên trái: $F[m, v - m]$.

Ví dụ $F[5, 5]$ sẽ được tính bằng $F[4, 5] + F[5, 0]$, hay $F[3, 5]$ sẽ được tính bằng $F[2, 5] + F[3, 2]$. Chính vì vậy để tính $F[m, v]$ thì $F[m - 1, v]$ và $F[m, v - m]$ phải được tính trước. Suy ra thứ tự hợp lý để tính các phần tử trong bảng F sẽ phải là theo thứ tự từ trên xuống và trên mỗi hàng thì tính theo thứ tự từ trái qua phải.

Điều đó có nghĩa là ban đầu ta phải tính hàng 0 của bảng: $F[0, v] =$ số dãy có các phần tử ≤ 0 mà tổng bằng v, theo quy ước ở đề bài thì $F[0, 0] = 1$ còn $F[0, v]$ với mọi $v > 0$ đều là 0.

Vậy giải thuật dựng rất đơn giản: Khởi tạo dòng 0 của bảng F: $F[0, 0] = 1$ còn $F[0, v]$ với mọi $v > 0$ đều bằng 0, sau đó dùng công thức truy hồi tính ra tất cả các phần tử của bảng F. Cuối cùng $F[n, n]$ là số cách phân tích cần tìm

```
P_3_01_1.PAS * Đếm số cách phân tích số n
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Analysel; {Bài toán phân tích số}
const
  max = 100;
var
  F: array[0..max, 0..max] of Integer;
  n, m, v: Integer;
begin
  Write('n = '); ReadLn(n);
  FillChar(F[0], SizeOf(F[0]), 0); {Khởi tạo dòng 0 của bảng F toàn số 0}
  F[0, 0] := 1; {Duy chỉ có F[0, 0] = 1}
  for m := 1 to n do {Dùng công thức tính các dòng theo thứ tự từ trên xuống dưới}
    for v := 0 to n do {Các phần tử trên một dòng thì tính theo thứ tự từ trái qua phải}
      if v < m then F[m, v] := F[m - 1, v]
      else F[m, v] := F[m - 1, v] + F[m, v - m];
  WriteLn(F[n, n], ' Analyses'); {Cuối cùng F[n, n] là số cách phân tích}
end.
```

1.2. CÁI TIẾN THỨ NHẤT

Cách làm trên có thể tóm tắt lại như sau: Khởi tạo dòng 0 của bảng, sau đó dùng dòng 0 tính dòng 1, dùng dòng 1 tính dòng 2 v.v... tới khi tính được hết dòng n. Có thể nhận thấy rằng khi đã tính xong dòng thứ k thì việc lưu trữ các dòng từ dòng 0 tới dòng k - 1 là không cần thiết bởi vì việc tính dòng k + 1 chỉ phụ thuộc các giá trị lưu trữ trên dòng k. Vậy ta có thể dùng hai mảng một chiều: Mảng Current lưu dòng hiện thời đang xét của bảng và mảng Next lưu dòng kế tiếp, đầu tiên mảng Current được gán các giá trị tương ứng trên dòng 0. Sau đó

dùng mảng Current tính mảng Next, mảng Next sau khi tính sẽ mang các giá trị tương ứng trên dòng 1. Rồi lại gán mảng Current := Next và tiếp tục dùng mảng Current tính mảng Next, mảng Next sẽ gồm các giá trị tương ứng trên dòng 2 v.v... Vậy ta có cài đặt cài tiến sau:

P_3_01_2.PAS * Đếm số cách phân tích số n

```
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Analyse2;
const
  max = 100;
var
  Current, Next: array[0..max] of Integer;
  n, m, v: Integer;
begin
  Write('n = '); ReadLn(n);
  FillChar(Current, SizeOf(Current), 0);
  Current[0] := 1; {Khởi tạo mảng Current tương ứng với dòng 0 của bảng F}
  for m := 1 to n do
    begin {Dùng dòng hiện thời Current tính dòng kế tiếp Next ⇔ Dùng dòng m - 1 tính dòng m của bảng F}
      for v := 0 to n do
        if v < m then Next[v] := Current[v]
        else Next[v] := Current[v] + Next[v - m];
      Current := Next; {Gán Current := Next tức là Current bây giờ lại lưu các phần tử trên dòng m của bảng F}
    end;
  WriteLn(Current[n], ' Analyses');
end.
```

Cách làm trên đã tiết kiệm được khá nhiều không gian lưu trữ, nhưng nó hơi chậm hơn phương pháp đầu tiên vì phép gán mảng (Current := Next). Có thể cài tiến thêm cách làm này như sau:

P_3_01_3.PAS * Đếm số cách phân tích số n

```
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Analyse3;
const
  max = 100;
var
  B: array[1..2, 0..max] of Integer;{Bảng B chỉ gồm 2 dòng thay cho 2 dòng liên tiếp của bảng phương án}
  n, m, v, x, y: Integer;
begin
  Write('n = '); ReadLn(n);
  {Trước hết, dòng 1 của bảng B tương ứng với dòng 0 của bảng phương án F, được điều chỉnh sẵn để có thể cộng dồn}
  FillChar(B[1], SizeOf(B[1]), 0);
  B[1][0] := 1;
  x := 1; {Đòng B[x] đóng vai trò là dòng hiện thời trong bảng phương án}
  y := 2; {Đòng B[y] đóng vai trò là dòng kế tiếp trong bảng phương án}
  for m := 1 to n do
    begin
      {Dùng dòng x tính dòng y ⇔ Dùng dòng hiện thời trong bảng phương án để tính dòng kế tiếp}
      for v := 0 to n do
        if v < m then B[y][v] := B[x][v]
        else B[y][v] := B[x][v] + B[y][v - m];
      x := 3 - x; y := 3 - y; {Đảo giá trị x và y, tính xoay lại}
    end;
  WriteLn(B[x][n], ' Analyses');
end.
```

1.3. CÀI TIẾN THỨ HAI

Ta vẫn còn cách tốt hơn nữa, tại mỗi bước, ta chỉ cần lưu lại một dòng của bảng F bằng một mảng 1 chiều, sau đó dùng mảng đó tính lại chính nó để sau khi tính, mảng một chiều sẽ lưu các giá trị của bảng F trên dòng kế tiếp.

```
P_3_01_4.PAS * Đếm số cách phân tích số n
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Analyse4;
const
  max = 100;
var
  L: array[0..max] of Integer; {Chỉ cần lưu 1 dòng}
  n, m, v: Integer;
begin
  Write('n = '); ReadLn(n);
  FillChar(L, SizeOf(L), 0);
  L[0] := 1; {Khởi tạo mảng 1 chiều L lưu dòng 0 của bảng}
  for m := 1 to n do {Dùng L tính lại chính nó}
    for v := m to n do
      L[v] := L[v] + L[v - m];
  WriteLn(L[n], ' Analyses');
end.
```

1.4. CÀI ĐẶT ĐỆ QUY

Xem lại công thức truy hồi tính $F[m, v] = F[m - 1, v] + F[m, v - m]$, ta nhận thấy rằng để tính $F[m, v]$ ta phải biết được chính xác $F[m - 1, v]$ và $F[m, v - m]$. Như vậy việc xác định thứ tự tính các phần tử trong bảng F (phần tử nào tính trước, phần tử nào tính sau) là quan trọng. Tuy nhiên ta có thể tính dựa trên một hàm đệ quy mà không cần phải quan tâm tới thứ tự tính toán.

Việc viết một hàm đệ quy tính công thức truy hồi khá đơn giản, như ví dụ này ta có thể viết:

```
P_3_01_5.PAS * Đếm số cách phân tích số n dùng đệ quy
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Analyse5;
var
  n: Integer;

function GetF(m, v: Integer): Integer;
begin
  if m = 0 then {Phần neo của hàm đệ quy}
    if v = 0 then GetF := 1
    else GetF := 0
  else {Phần đệ quy}
    if m > v then GetF := GetF(m - 1, v)
    else GetF := GetF(m - 1, v) + GetF(m, v - m);
  end;

  begin
    Write('n = '); ReadLn(n);
    WriteLn(GetF(n, n), ' Analyses');
  end.
```

Phương pháp cài đặt này tỏ ra khá chậm vì phải gọi nhiều lần mỗi hàm $GetF(m, v)$ (bài sau sẽ giải thích rõ hơn điều này). Ta có thể cài tiến bằng cách kết hợp với một mảng hai chiều F. Ban đầu các phần tử của F được coi là “chưa biết” (bằng cách gán một giá trị đặc biệt). Hàm $GetF(m, v)$ khi được gọi trước hết sẽ tra cứu tới $F[m, v]$, nếu $F[m, v]$ chưa biết thì hàm

GetF(m, v) sẽ gọi đệ quy để tính giá trị của F[m, v] rồi dùng giá trị này gán cho kết quả hàm, còn nếu F[m, v] đã biết thì hàm này chỉ việc gán kết quả hàm là F[m, v] mà không cần gọi đệ quy để tính toán nữa.

```
P_3_01_6.PAS * Đếm số cách phân tích số n dùng đệ quy
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Analyse6;
const
  max = 100;
var
  n: Integer;
  F: array[0..max, 0..max] of Integer;

function GetF(m, v: Integer): Integer;
begin
  if F[m, v] = -1 then {Nếu F[m, v] chưa biết thì đi tính F[m, v]}
    begin
      if m = 0 then {Phần neo của hàm đệ quy}
        if v = 0 then F[m, v] := 1
        else F[m, v] := 0
      else {Phần đệ quy}
        if m > v then F[m, v] := GetF(m - 1, v)
        else F[m, v] := GetF(m - 1, v) + GetF(m, v - m);
    end;
  GetF := F[m, v]; {Gán kết quả hàm bằng F[m, v]}
end;

begin
  Write('n = '); ReadLn(n);
  FillChar(f, SizeOf(f), $FF); {Khởi tạo mảng F bằng giá trị -1}
  WriteLn(GetF(n, n), ' Analyses');
end.
```

Việc sử dụng phương pháp đệ quy để giải công thức truy hồi là một kỹ thuật đáng lưu ý, vì khi gặp một công thức truy hồi phức tạp, khó xác định thứ tự tính toán thì phương pháp này tỏ ra rất hiệu quả, hơn thế nữa nó làm rõ hơn bản chất đệ quy của công thức truy hồi.

§2. PHƯƠNG PHÁP QUY HOẠCH ĐỘNG

2.1. BÀI TOÁN QUY HOẠCH

Bài toán quy hoạch là **bài toán tối ưu**: gồm có một hàm f gọi là hàm mục tiêu hay hàm đánh giá; các hàm g_1, g_2, \dots, g_n cho giá trị logic gọi là hàm ràng buộc. Yêu cầu của bài toán là tìm một cấu hình x thoả mãn tất cả các ràng buộc g_1, g_2, \dots, g_n : $g_i(x) = \text{TRUE}$ ($\forall i: 1 \leq i \leq n$) và x là tốt nhất, theo nghĩa không tồn tại một cấu hình y nào khác thoả mãn các hàm ràng buộc mà $f(y)$ tốt hơn $f(x)$.

Ví dụ:


Tìm (x, y) để

Hàm mục tiêu : $x + y \rightarrow \max$

Hàm ràng buộc : $x^2 + y^2 \leq 1$.

Xét trong mặt phẳng tọa độ, những cặp (x, y) thoả mãn $x^2 + y^2 \leq 1$ là tọa độ của những điểm nằm trong hình tròn có tâm O là gốc tọa độ, bán kính 1. Vậy nghiệm của bài toán bắt buộc nằm trong hình tròn đó.

Những đường thẳng có phương trình: $x + y = C$ (C là một hằng số) là đường thẳng vuông góc với đường phân giác góc phần tư thứ nhất. Ta phải tìm số C lớn nhất mà đường thẳng $x + y = C$ vẫn có điểm chung với đường tròn $(O, 1)$. Đường thẳng đó là một tiếp tuyến của đường tròn: $x + y = \sqrt{2}$. Tiếp điểm $\left(\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right)$ tương ứng với nghiệm tối ưu của bài toán đã cho.



Các dạng bài toán quy hoạch rất phong phú và đa dạng, ứng dụng nhiều trong thực tế, nhưng cũng cần biết rằng, đa số các bài toán quy hoạch là không giải được, hoặc chưa giải được. Cho đến nay, người ta mới chỉ có thuật toán đơn hình giải bài toán quy hoạch tuyến tính lồi, và một vài thuật toán khác áp dụng cho các lớp bài toán cụ thể.

2.2. PHƯƠNG PHÁP QUY HOẠCH ĐỘNG

Phương pháp quy hoạch động dùng để giải bài toán tối ưu có bản chất đê quy, tức là việc tìm phương án tối ưu cho bài toán đó có thể đưa về tìm phương án tối ưu của một số hữu hạn các bài toán con. Đối với nhiều thuật toán đê quy chúng ta đã tìm hiểu, nguyên lý chia để trị (divide and conquer) thường đóng vai trò chủ đạo trong việc thiết kế thuật toán. Để giải quyết

một bài toán lớn, ta chia nó làm nhiều bài toán con cùng dạng với nó để có thể giải quyết độc lập. Trong phương pháp quy hoạch động, nguyên lý này càng được thể hiện rõ: Khi không biết cần phải giải quyết những bài toán con nào, ta sẽ đi giải quyết tất cả các bài toán con và lưu trữ những lời giải hay đáp số của chúng với mục đích sử dụng lại theo một sự phối hợp nào đó để giải quyết những bài toán tổng quát hơn. Đó chính là điểm khác nhau giữa Quy hoạch động và phép phân giải đệ quy và cũng là nội dung phương pháp quy hoạch động:

- ❖ Phép phân giải đệ quy bắt đầu từ bài toán lớn phân rã thành nhiều bài toán con và đi giải từng bài toán con đó. Việc giải từng bài toán con lại đưa về phép phân rã tiếp thành nhiều bài toán nhỏ hơn và lại đi giải tiếp bài toán nhỏ hơn đó bắt kể nó đã được giải hay chưa.
- ❖ Quy hoạch động bắt đầu từ việc giải tất cả các bài toán nhỏ nhất (bài toán cơ sở) để từ đó từng bước giải quyết những bài toán lớn hơn, cho tới khi giải được bài toán lớn nhất (bài toán ban đầu).

Ta xét một ví dụ đơn giản:

Dãy Fibonacci là dãy vô hạn các số nguyên dương $F[1], F[2], \dots$ được định nghĩa như sau:

$$F[i] = \begin{cases} 1, & \text{if } i \leq 2 \\ F[i-1] + F[i-2], & \text{if } i \geq 3 \end{cases}$$

Hãy tính $F[6]$

Xét hai cách cài đặt chương trình:


Cách 1

```
program Fibol;
function F(i: Integer): Integer;
begin
  if i < 3 then F := 1
  else F := F(i - 1) + F(i - 2);
end;
begin
  WriteLn(F(6));
end.
```

Cách 2

```
program Fibo2;
var
  F: array[1..6] of Integer;
  i: Integer;
begin
  F[1] := 1; F[2] := 1;
  for i := 3 to 6 do
    F[i] := F[i - 1] + F[i - 2];
  WriteLn(F[6]);
end.
```

Cách 1 có hàm đệ quy $F(i)$ để tính số Fibonacci thứ i . Chương trình chính gọi $F(6)$, nó sẽ gọi tiếp $F(5)$ và $F(4)$ để tính ... Quá trình tính toán có thể vẽ như cây dưới đây. Ta nhận thấy để tính $F(6)$ nó phải tính 1 lần $F(5)$, hai lần $F(4)$, ba lần $F(3)$, năm lần $F(2)$, ba lần $F(1)$.

**Hình 49: Hàm đệ quy tính số Fibonacci**

Cách 2 thì không như vậy. Trước hết nó tính sẵn $F[1]$ và $F[2]$, từ đó tính tiếp $F[3]$, lại tính tiếp được $F[4]$, $F[5]$, $F[6]$. Đảm bảo rằng mỗi giá trị Fibonacci chỉ phải tính 1 lần.

(Cách 2 còn có thể cải tiến thêm nữa, chỉ cần dùng 3 giá trị tính lặp nhau)

Trước khi áp dụng phương pháp quy hoạch động ta phải xét xem phương pháp đó có thoả mãn những yêu cầu dưới đây hay không:

- ❖ Bài toán lớn phải phân rã được thành nhiều bài toán con, mà sự phối hợp lời giải của các bài toán con đó cho ta lời giải của bài toán lớn.
- ❖ Vì quy hoạch động là đi giải tất cả các bài toán con, nên nếu không đủ không gian vật lý lưu trữ lời giải (bộ nhớ, đĩa...) để phối hợp chúng thì phương pháp quy hoạch động cũng không thể thực hiện được.
- ❖ Quá trình từ bài toán cơ sở tìm ra lời giải bài toán ban đầu phải qua hữu hạn bước.

Các khái niệm:

- ❖ Bài toán giải theo phương pháp quy hoạch động gọi là **bài toán quy hoạch động**
- ❖ Công thức phối hợp nghiệm của các bài toán con để có nghiệm của bài toán lớn gọi là **công thức truy hồi** (hay phương trình truy toán) của quy hoạch động
- ❖ Tập các bài toán nhỏ nhất có ngay lời giải để từ đó giải quyết các bài toán lớn hơn gọi là **cơ sở quy hoạch động**
- ❖ Không gian lưu trữ lời giải các bài toán con để tìm cách phối hợp chúng gọi là **bảng phương án của quy hoạch động**

Các bước cài đặt một chương trình sử dụng quy hoạch động:

- ❖ Giải tất cả các bài toán cơ sở (thông thường rất dễ), lưu các lời giải vào bảng phương án.
- ❖ Dùng công thức truy hồi phối hợp những lời giải của những bài toán nhỏ đã lưu trong bảng phương án để tìm lời giải của những bài toán lớn hơn và lưu chúng vào bảng phương án. Cho tới khi bài toán ban đầu tìm được lời giải.
- ❖ Dựa vào bảng phương án, truy vết tìm ra nghiệm tối ưu.

Cho đến nay, vẫn chưa có một định lý nào cho biết một cách chính xác những bài toán nào có thể giải quyết hiệu quả bằng quy hoạch động. Tuy nhiên để biết được bài toán có thể giải bằng quy hoạch động hay không, ta có thể tự đặt câu hỏi: "**Một nghiệm tối ưu của bài toán lớn có phải là sự phối hợp các nghiệm tối ưu của các bài toán con hay không?**" và "**Liệu có thể nào lưu trữ được nghiệm các bài toán con dưới một hình thức nào đó để phối hợp tìm được nghiệm bài toán lớn?**"

§3. MỘT SỐ BÀI TOÁN QUY HOẠCH ĐỘNG

3.1. DÃY CON ĐƠN ĐIỆU TĂNG DÀI NHẤT

Cho dãy số nguyên $A = a[1..n]$. ($n \leq 10^6$, $-10^6 \leq a[i] \leq 10^6$). Một dãy con của A là một cách chọn ra trong A một số phần tử giữ nguyên thứ tự. Như vậy A có 2^n dãy con.

Yêu cầu: Tìm dãy con đơn điệu tăng của A có độ dài lớn nhất.

Ví dụ: $A = (1, 2, 3, 4, 9, 10, 5, 6, 7)$. Dãy con đơn điệu tăng dài nhất là: $(1, 2, 3, 4, 5, 6, 7)$.

Input: file văn bản INCSEQ.INP

- ❖ Dòng 1: Chứa số n
- ❖ Dòng 2: Chứa n số $a[1], a[2], \dots, a[n]$ cách nhau ít nhất một dấu cách

Output: file văn bản INCSEQ.OUT

- ❖ Dòng 1: Ghi độ dài dãy con tìm được
- ❖ Các dòng tiếp: ghi dãy con tìm được và chỉ số những phần tử được chọn vào dãy con đó.

INCSEQ.INP	INCSEQ.OUT
11	8
1 2 3 8 9 4 5 6 20 9 10	$a[1] = 1$
	$a[2] = 2$
	$a[3] = 3$
	$a[6] = 4$
	$a[7] = 5$
	$a[8] = 6$
	$a[10] = 9$
	$a[11] = 10$

Cách giải:

Bổ sung vào A hai phần tử: $a[0] = -\infty$ và $a[n+1] = +\infty$. **Khi đó dãy con đơn điệu tăng dài nhất chắc chắn sẽ bắt đầu từ $a[0]$ và kết thúc ở $a[n+1]$.**

Với $\forall i: 0 \leq i \leq n + 1$. Ta sẽ tính $L[i] =$ độ dài dãy con đơn điệu tăng dài nhất bắt đầu tại $a[i]$.

3.1.1. Cơ sở quy hoạch động (bài toán nhỏ nhất):

$L[n+1] =$ Độ dài dãy con đơn điệu tăng dài nhất bắt đầu tại $a[n+1] = +\infty$. Dãy con này chỉ gồm mỗi một phần tử ($+\infty$) nên $L[n+1] = 1$.

3.1.2. Công thức truy hồi:

Giả sử với i chạy từ n về 0 , ta cần tính $L[i]$: độ dài dãy con tăng dài nhất bắt đầu tại $a[i]$. $L[i]$ được tính trong điều kiện $L[i+1..n+1]$ đã biết:

Dãy con đơn điệu tăng dài nhất bắt đầu từ $a[i]$ sẽ được thành lập bằng cách lấy $a[i]$ ghép vào đầu một trong số những dãy con đơn điệu tăng dài nhất bắt đầu tại vị trí $a[j]$ đứng sau $a[i]$. Ta sẽ chọn dãy nào để ghép $a[i]$ vào đầu? Tất nhiên là chỉ được ghép $a[i]$ vào đầu những dãy con bắt đầu tại $a[j]$ nào đó lớn hơn $a[i]$ (để đảm bảo tính tăng) và dĩ nhiên ta sẽ chọn dãy dài nhất để ghép $a[i]$ vào đầu (để đảm bảo tính dài nhất). Vậy $L[i]$ được tính như sau: **Xét tất cả các**

chỉ số j trong khoảng từ i + 1 đến n + 1 mà $a[j] > a[i]$, chọn ra chỉ số jmax có $L[jmax]$ lớn nhất. Đặt $L[i] := L[jmax] + 1$:

$$L[i] = \max_{\substack{i < j \leq n-1 \\ a[i] < a[j]}} L[j] + 1$$

3.1.3. Truy vết

Tại bước xây dựng dãy L, mỗi khi gán $L[i] := L[jmax] + 1$, ta đặt $T[i] = jmax$. Để lưu lại rằng: Dãy con dài nhất bắt đầu tại $a[i]$ sẽ có phần tử thứ hai kế tiếp là $a[jmax]$.

Sau khi tính xong hay dãy L và T, ta bắt đầu từ $T[0]$.

$T[0]$ chính là phần tử đầu tiên được chọn,

$T[T[0]]$ là phần tử thứ hai được chọn,

$T[T[T[0]]]$ là phần tử thứ ba được chọn ...

Quá trình truy vết có thể diễn tả như sau:

```
i := T[0];
while i < n + 1 do {Chừng nào chưa duyệt đến số a[n+1]=+∞ ở cuối}
begin
  <Thông báo chọn a[i]>
  i := T[i];
end;
```

Ví dụ: với $A = (5, 2, 3, 4, 9, 10, 5, 6, 7, 8)$. Hai dãy L và T sau khi tính sẽ là:

Calculating												
i	0	1	2	3	4	5	6	7	8	9	10	11
a_i	$-\infty$	5	2	3	4	9	10	5	6	7	8	$+\infty$
$L[i]$	9	5	8	7	6	3	2	5	4	3	2	1
$T[i]$	2	8	3	4	7	6	11	8	9	10	11	

Tracing

Hình 50: Tính toán và truy vết

```
P_3_03_1.PAS * Tìm dãy con đơn điệu tăng dài nhất
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program LongestSubSequence;
const
  InputFile = 'INCSEQ.INP';
  OutputFile = 'INCSEQ.OUT';
  max = 1000000;
var
  a, L, T: array[0..max + 1] of Integer;
  n: Integer;

procedure Enter;
var
  i: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n);
```

```

for i := 1 to n do Read(f, a[i]);
Close(f);
end;

procedure Optimize; {Quy hoạch động}
var
  i, j, jmax: Integer;
begin
  a[0] := Low(Integer); a[n + 1] := High(Integer); {Thêm hai phần tử cạnh hai đầu dãy a}
  L[n + 1] := 1; {Điền cơ sở quy hoạch động vào bảng phương án}
  for i := n downto 0 do {Tính bảng phương án}
    begin
      {Chọn trong các chỉ số j đúng sau i thoả mãn a[j] > a[i] ra chỉ số jmax có L[jmax] lớn nhất}
      jmax := n + 1;
      for j := i + 1 to n + 1 do
        if (a[j] > a[i]) and (L[j] > L[jmax]) then jmax := j;
      L[i] := L[jmax] + 1; {Lưu độ dài dãy con tăng dài nhất bắt đầu tại a[i]}
      T[i] := jmax; {Lưu vết: phần tử đúng liền sau a[i] trong dãy con tăng dài nhất đó là a[jmax]}
    end;
  end;

procedure Result;
var
  f: Text;
  i: Integer;
begin
  Assign(f, OutputFile); Rewrite(f);
  WriteLn(f, L[0] - 2); {Chiều dài dãy con tăng dài nhất}
  i := T[0]; {Bắt đầu truy vết tìm nghiệm}
  while i <> n + 1 do
    begin
      WriteLn(f, 'a[' , i, '] = ', a[i]);
      i := T[i];
    end;
  Close(f);
end;

begin
  Enter;
  Optimize;
  Result;
end.

```

Nhận xét:

Nhắc lại công thức truy hồi tính các L[.] là:

$$\begin{cases} L[n+1] = 0 \\ L[i] = \max_{\substack{i < j \leq n+1 \\ a[i] < a[j]}} L[j] + 1; (\forall i=0,n) \end{cases}$$

và để tính hết các L[.], ta phải mất một đoạn chương trình với độ phức tạp tính toán là O(n²).

Ta có thể cải tiến cách cài đặt để được một đoạn chương trình với độ phức tạp tính toán là O(nlogn) bằng kỹ thuật sau:

Với mỗi số k, ta gọi StartOf[k] là chỉ số x của phần tử a[x] thoả mãn: dãy đơn điệu tăng dài nhất bắt đầu từ a[x] có độ dài k. Nếu có nhiều phần tử a[.] cùng thoả mãn điều kiện này thì ta chọn phần tử a[x] là phần tử lớn nhất trong số những phần tử đó. Việc tính các giá trị StartOf[.] được thực hiện đồng thời với việc tính các giá trị L[.] bằng phương pháp sau:

```

L[n + 1] := 1;
StartOf[1] := n + 1;
m := 1; {m là độ dài dãy con đơn điệu tăng dài nhất của dãy a[i..n+1] (ở bước khởi tạo này i = n + 1)}
for i := n downto 0 do
begin
  {Tính L[i]; đặt k := L[i];}
  if k > m then {Nếu dãy con tăng dài nhất bắt đầu tại a[i] có độ dài > m}
    begin
      m := k; {Cập nhật lại m}
      StartOf[k] := i; {Gán giá trị cho StartOf[m]}
    end
  else
    if a[i] > a[StartOf[k]] then {Nếu có nhiều dãy đơn điệu tăng dài nhất độ dài k thì}
      StartOf[k] := i; {ghi nhận lại dãy có phần tử bắt đầu lớn nhất}
  end;

```

3.1.4. Cải tiến

Khi bắt đầu vào một lần lặp với một giá trị i, ta đã biết được:

- ❖ m: Độ dài dãy con đơn điệu tăng dài nhất của dãy a[i+1..n+1]
- ❖ StartOf[k] ($1 \leq k \leq m$): Phần tử a[StartOf[k]] là phần tử lớn nhất trong số các phần tử trong đoạn a[i+1..n+1] thoả mãn: Dãy con đơn điệu tăng dài nhất bắt đầu từ a[StartOf[k]] có độ dài k. Do thứ tự tính toán được áp đặt như trong sơ đồ trên, ta dễ dàng nhận thấy rằng: $a[StartOf[k]] < a[StartOf[k - 1]] < \dots < a[StartOf[1]]$.

Điều kiện để có dãy con đơn điệu tăng độ dài p+1 bắt đầu tại a[i] chính là $a[StartOf[p]] > a[i]$ (vì theo thứ tự tính toán thì khi bắt đầu một lần lặp với giá trị i, a[StartOf[p]] luôn đứng sau a[i]). Mặt khác nếu đem a[i] ghép vào đầu dãy con đơn điệu tăng dài nhất bắt đầu tại a[StartOf[p]] mà thu được dãy tăng thì đem a[i] ghép vào đầu dãy con đơn điệu tăng dài nhất bắt đầu tại a[StartOf[p - 1]] ta cũng thu được dãy tăng. Vậy để tính L[i], ta có thể tìm số p lớn nhất thoả mãn $a[StartOf[p]] > a[i]$ bằng **thuật toán tìm kiếm nhị phân** rồi đặt L[i] := p + 1 (và sau đó T[i] := StartOf[p], tất nhiên)

P_3_03_2.PAS * Cải tiến thuật toán tìm dãy con đơn điệu tăng dài nhất

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program LongestSubSequence;
const
  InputFile = 'INCSEQ.INP';
  OutputFile = 'INCSEQ.OUT';
const
  max = 1000000;
var
  a, L, T, StartOf: array[0..max + 1] of Integer;
  n, m: Integer;

procedure Enter;
var
  i: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n);
  for i := 1 to n do Read(f, a[i]);
  Close(f);
end;

procedure Init;

```

```

begin
  a[0] := Low(Integer);
  a[n + 1] := High(Integer);
  m := 1;
  L[n + 1] := 1;
  StartOf[1] := n + 1;
end;

{Hàm Find, tìm vị trí j mà nếu đem ai ghép vào đầu dãy con đơn điệu tăng dài nhất bắt đầu từ a_j sẽ được dãy đơn
điệu tăng dài nhất bắt đầu tại a_i}
function Find(i: Integer): Integer;
var
  inf, sup, median, j: Integer;
begin
  inf := 1; sup := m + 1;
  repeat {Thuật toán tìm kiếm nhị phân}
    median := (inf + sup) div 2;
    j := StartOf[median];
    if a[j] > a[i] then inf := median {Luôn để aStartOf[inf] > ai ≥ aStartOf[sup]}
    else sup := median;
  until inf + 1 = sup;
  Find := StartOf[inf];
end;

procedure Optimize;
var
  i, j, k: Integer;
begin
  for i := n downto 0 do
    begin
      j := Find(i);
      k := L[j] + 1;
      if k > m then
        begin
          m := k;
          StartOf[k] := i;
        end
      else
        if a[StartOf[k]] < a[i] then
          StartOf[k] := i;
      L[i] := k;
      T[i] := j;
    end;
end;

procedure Result;
var
  f: Text;
  i: Integer;
begin
  Assign(f, OutputFile); Rewrite(f);
  WriteLn(f, m - 2);
  i := T[0];
  while i <> n + 1 do
    begin
      WriteLn(f, 'a['', i, ''] = ', a[i]);
      i := T[i];
    end;
  Close(f);
end;

begin
  Enter;

```

```

Init;
Optimize;
Result;
end.

```

Dễ thấy chi phí thời gian thực hiện giải thuật này cấp $O(n\log n)$, đây là một ví dụ điển hình cho thấy rằng một công thức truy hồi có thể có nhiều phương pháp tính.

3.2. BÀI TOÁN CÁI TÚI

Trong siêu thị có n gói hàng ($n \leq 100$), gói hàng thứ i có trọng lượng là $W[i] \leq 100$ và trị giá $V[i] \leq 100$. Một tên trộm đột nhập vào siêu thị, tên trộm mang theo một cái túi có thể mang được tối đa trọng lượng M ($M \leq 100$). Hỏi tên trộm sẽ lấy đi những gói hàng nào để được tổng giá trị lớn nhất.

Input: file văn bản BAG.INP

- ❖ Dòng 1: Chứa hai số n, M cách nhau ít nhất một dấu cách
- ❖ n dòng tiếp theo, dòng thứ i chứa hai số nguyên dương $W[i], V[i]$ cách nhau ít nhất một dấu cách

Output: file văn bản BAG.OUT

- ❖ Dòng 1: Ghi giá trị lớn nhất tên trộm có thể lấy
- ❖ Dòng 2: Ghi chỉ số những gói bị lấy

BAG.INP	BAG.OUT
5 11	11
3 3	5 2 1
4 4	
5 4	
9 10	
4 4	

Cách giải:

Nếu gọi $F[i, j]$ là giá trị lớn nhất có thể có bằng cách chọn trong các gói $\{1, 2, \dots, i\}$ với giới hạn trọng lượng j . Thì giá trị lớn nhất khi được chọn trong số n gói với giới hạn trọng lượng M chính là $F[n, M]$.

3.2.1. Công thức truy hồi tính $F[i, j]$.

Với giới hạn trọng lượng j , việc chọn tối ưu trong số các gói $\{1, 2, \dots, i - 1, i\}$ để có giá trị lớn nhất sẽ có hai khả năng:

- ❖ Nếu không chọn gói thứ i thì $F[i, j]$ là giá trị lớn nhất có thể bằng cách chọn trong số các gói $\{1, 2, \dots, i - 1\}$ với giới hạn trọng lượng là j . Tức là

$$F[i, j] = F[i - 1, j]$$

- ❖ Nếu có chọn gói thứ i (tất nhiên chỉ xét tới trường hợp này khi mà $W[i] \leq j$) thì $F[i, j]$ bằng giá trị gói thứ i là $V[i]$ cộng với giá trị lớn nhất có thể có được bằng cách chọn trong số các gói $\{1, 2, \dots, i - 1\}$ với giới hạn trọng lượng $j - W[i]$. Tức là về mặt giá trị thu được:

$$F[i, j] = V[i] + F[i - 1, j - W[i]]$$

Vì theo cách xây dựng $F[i, j]$ là giá trị lớn nhất có thể, nên $F[i, j]$ sẽ là max trong 2 giá trị thu được ở trên.

3.2.2. Cơ sở quy hoạch động:

Dễ thấy $F[0, j] =$ giá trị lớn nhất có thể bằng cách chọn trong số 0 gói = 0.

3.2.3. Tính bảng phương án:

Bảng phương án F gồm $n + 1$ dòng, $M + 1$ cột, trước tiên được điền cơ sở quy hoạch động: Dòng 0 gồm toàn số 0. Sử dụng công thức truy hồi, dùng dòng 0 tính dòng 1, dùng dòng 1 tính dòng 2, v.v... đến khi tính hết dòng n .

F	0	1	2	M	
0	0	0	0	...0...	0	
1						
2						
...	
n						

3.2.4. Truy vết:

Tính xong bảng phương án thì ta quan tâm đến $F[n, M]$ đó chính là giá trị lớn nhất thu được khi chọn trong cả n gói với giới hạn trọng lượng M . Nếu $F[n, M] = F[n - 1, M]$ thì tức là không chọn gói thứ n , ta truy tiếp $F[n - 1, M]$. Còn nếu $F[n, M] \neq F[n - 1, M]$ thì ta thông báo rằng phép chọn tối ưu có chọn gói thứ n và truy tiếp $F[n - 1, M - W[n]]$. Cứ tiếp tục cho đến khi truy lên tới hàng 0 của bảng phương án.

```
P_3_03_3.PAS * Bài toán cái túi
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program The_Bag;
const
  InputFile = 'BAG.INP';
  OutputFile = 'BAG.OUT';
  max = 100;
var
  W, V: Array[1..max] of Integer;
  F: array[0..max, 0..max] of Integer;
  n, M: Integer;

procedure Enter;
var
  i: Integer;
  fi: Text;
begin
  Assign(fi, InputFile); Reset(fi);
  ReadLn(fi, n, M);
  for i := 1 to n do ReadLn(fi, W[i], V[i]);
  Close(fi);
end;

procedure Optimize; {Tính bảng phương án bằng công thức truy hồi}
var
  i, j: Integer;
begin
```

```

FillChar(F[0], SizeOf(F[0]), 0); {Điền cơ sở quy hoạch động}
for i := 1 to n do
  for j := 0 to M do
    begin {Tính F[i, j]}
      F[i, j] := F[i - 1, j]; {Giả sử không chọn gói thú i thì F[i, j] = F[i - 1, j]}
      {Sau đó đánh giá: nếu chọn gói thú i sẽ được lợi hơn thì đặt lại F[i, j]}
      if (j >= W[i]) and (F[i, j] < F[i - 1, j - W[i]] + V[i]) then
        F[i, j] := F[i - 1, j - W[i]] + V[i];
    end;
  end;

procedure Trace; {Truy vết tìm nghiệm tối ưu}
var
  fo: Text;
begin
  Assign(fo, OutputFile); Rewrite(fo);
  WriteLn(fo, F[n, M]); {In ra giá trị lớn nhất có thể kiếm được}
  while n <> 0 do {Truy vết trên bảng phương án từ hàng n lên hàng 0}
    begin
      if F[n, M] <> F[n - 1, M] then {Nếu có chọn gói thứ n}
        begin
          Write(fo, n, ' ');
          M := M - W[n]; {Đã chọn gói thứ n rồi thì chỉ có thể mang thêm được trọng lượng M - W[n] nữa thôi}
        end;
      Dec(n);
    end;
  Close(fo);
end;

begin
  Enter;
  Optimize;
  Trace;
end.

```

3.3. BIÊN ĐỔI XÂU

Cho xâu ký tự X, xét 3 phép biến đổi:

- a) Insert(i, C): i là số, C là ký tự: Phép Insert chèn ký tự C vào sau vị trí i của xâu X.
- b) Replace(i, C): i là số, C là ký tự: Phép Replace thay ký tự tại vị trí i của xâu X bởi ký tự C.
- c) Delete(i): i là số, Phép Delete xoá ký tự tại vị trí i của xâu X.

Yêu cầu: Cho trước xâu Y, hãy tìm một số ít nhất các phép biến đổi trên để biến xâu X thành xâu Y.

Input: file văn bản STR.INP

- ❖ Dòng 1: Chứa xâu X (độ dài ≤ 100)
- ❖ Dòng 2: Chứa xâu Y (độ dài ≤ 100)

Output: file văn bản STR.OUT ghi các phép biến đổi cần thực hiện và xâu X tại mỗi phép biến đổi.

STR. INP	STR. OUT
PBBCEFATZ	7
QABCDABEFA	PBBCEFATZ -> Delete(9) -> PBBCEFAT PBBCEFAT -> Delete(8) -> PBBCEFA PBBCEFA -> Insert(4, B) -> PBBCBEFA PBBCBEFA -> Insert(4, A) -> PBBCABEFA PBBCABEFA -> Insert(4, D) -> PBBCDABEFA BBCDABEFA -> Replace(2, A) -> PABCDABEFA PABCDABEFA -> Replace(1, Q) -> QABCDABEFA

Cách giải:

Đối với xâu ký tự thì việc xoá, chèn sẽ làm cho các phần tử phía sau vị trí biến đổi bị đánh chỉ số lại, gây khó khăn cho việc quản lý vị trí. Để khắc phục điều này, ta sẽ tìm một thứ tự biến đổi thoả mãn: Phép biến đổi tại vị trí i bắt buộc phải thực hiện sau các phép biến đổi tại vị trí $i + 1, i + 2, \dots$

Ví dụ: $X = 'ABCD'$;

Insert(0, E) sau đó Delete(4) cho ra $X = 'EABD'$. Cách này không tuân thủ nguyên tắc Delete(3) sau đó Insert(0, E) cho ra $X = 'EABD'$. Cách này tuân thủ nguyên tắc đè ra.

Nói tóm lại ta sẽ tìm một dãy biến đổi có vị trí thực hiện giảm dần.

3.3.1. Công thức truy hồi

Giả sử m là độ dài xâu X và n là độ dài xâu Y . Gọi $F[i, j]$ là số phép biến đổi tối thiểu để biến xâu gồm i ký tự đầu của xâu X : $X[1..i]$ thành xâu gồm j ký tự đầu của xâu Y : $Y[1..j]$.

Quan sát hai dãy X và Y




Ta nhận thấy:

❖ Nếu $X[m] = Y[n]$ thì ta chỉ cần biến đoạn $X[1..m-1]$ thành $Y[1..n-1]$. Tức là trong trường hợp này: $F[m, n] = F[m - 1, n - 1]$



❖ Nếu $X[m] \neq Y[n]$ thì tại vị trí $X[m]$ ta có thể sử dụng một trong 3 phép biến đổi:

Hoặc chèn vào sau vị trí m của X , một ký tự đúng bằng Y_n :




Thì khi đó $F[m, n]$ sẽ bằng 1 phép chèn vừa rồi cộng với số phép biến đổi biến dãy $X[1..m]$ thành dãy $Y[1..n-1]$: $F[m, n] = 1 + F[m, n - 1]$

Hoặc thay vị trí m của X bằng một ký tự đúng bằng Y[n]:



Thì khi đó $F[m, n]$ sẽ bằng 1 phép thay vừa rồi cộng với số phép biến đổi biến dãy $X[1..m-1]$ thành dãy $Y[1..n-1]$: $F[m, n] = 1 + F[m-1, n-1]$

Hoặc xoá vị trí thứ m của X:



Thì khi đó $F[m, n]$ sẽ bằng 1 phép xoá vừa rồi cộng với số phép biến đổi biến dãy $X[1..m-1]$ thành dãy $Y[1..n]$: $F[m, n] = 1 + F[m-1, n]$

Vì $F[m, n]$ phải là nhỏ nhất có thể, nên trong trường hợp $X[m] \neq Y[n]$ thì

$$F[m, n] = \min(F[m, n - 1], F[m - 1, n - 1], F[m - 1, n]) + 1.$$

Ta xây dựng xong công thức truy hồi:

$$F[m, n] = \begin{cases} F[m-1, n-1], & \text{if } X_m = Y_n \\ \min(F[m, n-1], F[m-1, n-1], F[m-1, n]) + 1, & \text{if } X_m \neq Y_n \end{cases}$$

3.3.2. Cơ sở quy hoạch động

- ❖ $F[0, j]$ là số phép biến đổi biến xâu rỗng thành xâu gồm j ký tự đầu của F. Nó cần tối thiểu j phép chèn: $F[0, j] = j$
- ❖ $F[i, 0]$ là số phép biến đổi biến xâu gồm i ký tự đầu của S thành xâu rỗng, nó cần tối thiểu i phép xoá: $F[i, 0] = i$

Vậy đầu tiên bảng phương án F ($cô[0..m, 0..n]$) được khởi tạo hàng 0 và cột 0 là cơ sở quy hoạch động. Từ đó dùng công thức truy hồi tính ra tất cả các phần tử bảng B.

Sau khi tính xong thì $F[m, n]$ cho ta biết số phép biến đổi tối thiểu.

Truy vết:

Nếu $X[m] = Y[n]$ thì chỉ việc xét tiếp $F[m - 1, n - 1]$.

Nếu không, xét 3 trường hợp:

- ❖ Nếu $F[m, n] = F[m, n - 1] + 1$ thì phép biến đổi đầu tiên được sử dụng là: Insert(m, Y[n])
- ❖ Nếu $F[m, n] = F[m - 1, n - 1] + 1$ thì phép biến đổi đầu tiên được sử dụng là: Replace(m, Y[n])
- ❖ Nếu $F[m, n] = F[m - 1, n] + 1$ thì phép biến đổi đầu tiên được sử dụng là: Delete(m)

Đưa về bài toán với m, n nhỏ hơn truy vết tiếp cho tới khi về $F[0, 0]$

Ví dụ: X = 'ABCD'; Y = 'EABD' bảng phương án là:

F	0	1	2	3	4
0	0	1	2	3	4
1	1	1	1	2	3
2	2	2	2	1	2
3	3	3	3	2	2
4	4	4	4	3	2

Hình 51: Truy vết

Lưu ý: khi truy vết, để tránh truy nhập ra ngoài bảng, nên tạo viền cho bảng.

P_3_03_4.PAS * Biến đổi xâu

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program StrOpt;
const
  InputFile = 'STR.INP';
  OutputFile = 'STR.OUT';
  max = 100;
var
  X, Y: String[2 * max];
  F: array[-1..max, -1..max] of Integer;
  m, n: Integer;

procedure Enter;
var
  fi: Text;
begin
  Assign(fi, InputFile); Reset(fi);
  ReadLn(fi, X); ReadLn(fi, Y);
  Close(fi);
  m := Length(X); n := Length(Y);
end;

function Min3(x, y, z: Integer): Integer; {Cho giá trị nhỏ nhất trong 3 giá trị x, y, z}
var
  t: Integer;
begin
  if x < y then t := x else t := y;
  if z < t then t := z;
  Min3 := t;
end;

procedure Optimize;
var
  i, j: Integer;
begin
  {Khởi tạo viền cho bảng phuong án}
  for i := 0 to m do F[i, -1] := max + 1;
  for j := 0 to n do F[-1, j] := max + 1;
  {Lưu cơ sở quy hoạch động}
  for j := 0 to n do F[0, j] := j;
  for i := 1 to m do F[i, 0] := i;
  {Dùng công thức truy hồi tính toàn bảng phuong án}
  for i := 1 to m do
    for j := 1 to n do
      if X[i] = Y[j] then F[i, j] := F[i - 1, j - 1]
      else F[i, j] := Min3(F[i, j - 1], F[i - 1, j - 1], F[i - 1, j]) + 1;
end;

```

```

end;

procedure Trace; {Truy vết}
var
  fo: Text;
begin
  Assign(fo, OutputFile); Rewrite(fo);
  WriteLn(fo, F[m, n]); {F[m, n] chính là số ít nhất các phép biến đổi cần thực hiện}
  while (m <> 0) or (n <> 0) do {Vòng lặp kết thúc khi m = n = 0}
    if X[m] = Y[n] then {Hai ký tự cuối của 2 xâu giống nhau}
      begin
        Dec(m); Dec(n); {Chi việc truy chéo lên trên bảng phương án}
      end
    else {Tại đây cần một phép biến đổi}
      begin
        Write(fo, X, ' -> '); {In ra xâu X trước khi biến đổi}
        if F[m, n] = F[m, n - 1] + 1 then {Nếu đây là phép chèn}
          begin
            Write(fo, 'Insert('' , m, '' , '' , Y[n], '')');
            Insert(Y[n], X, m + 1);
            Dec(n); {Truy sang phải}
          end
        else
          if F[m, n] = F[m - 1, n - 1] + 1 then {Nếu đây là phép thay}
            begin
              Write(fo, 'Replace('' , m, '' , '' , Y[n], '')');
              X[m] := Y[n];
              Dec(m); Dec(n); {Truy chéo lên trên}
            end
          else {Nếu đây là phép xoá}
            begin
              Write(fo, 'Delete('' , m, '')');
              Delete(X, m, 1);
              Dec(m); {Truy lên trên}
            end;
        WriteLn(fo, ' -> ', X); {In ra xâu X sau phép biến đổi}
      end;
  Close(fo);
end;

begin
  Enter;
  Optimize;
  Trace;
end.

```

Hãy tự giải thích tại sao khi giới hạn độ dài dữ liệu là 100, lại phải khai báo X và Y là String[200] chứ không phải là String[100] ?.

3.4. DÃY CON CÓ TỔNG CHIA HẾT CHO K

Cho một dãy A gồm n ($1 \leq n \leq 1000$) số nguyên dương $a[1..n]$ và số nguyên dương k ($k \leq 1000$). Hãy tìm dãy con gồm nhiều phần tử nhất của dãy đã cho sao cho tổng các phần tử của dãy con này chia hết cho k.

Input: file văn bản SUBSEQ.INP

- ❖ Dòng 1: Chứa số n
- ❖ Dòng 2: Chứa n số $a[1], a[2], \dots, a[n]$ cách nhau ít nhất một dấu cách

Output: file văn bản SUBSEQ.OUT

- ❖ Dòng 1: Ghi độ dài dãy con tìm được
- ❖ Các dòng tiếp: Ghi các phần tử được chọn vào dãy con
- ❖ Dòng cuối: Ghi tổng các phần tử của dãy con đó.

SUBSEQ.INP	SUBSEQ.OUT
10 5	8
1 6 11 5 10 15 20 2 4 9	a[10] = 9 a[9] = 4 a[7] = 20 a[6] = 15 a[5] = 10 a[4] = 5 a[3] = 11 a[2] = 6 Sum = 80

3.4.1. Cách giải 1

Không ảnh hưởng đến kết quả cuối cùng, ta có thể đặt: $a[i] := a[i] \bmod k$ với $\forall i: 1 \leq i \leq n$. Gọi S là tổng các phần tử trong dãy A , thay đổi cách tiếp cận bài toán: thay vì tìm xem phải chọn ra một số tối đa những phần tử để có tổng chia hết cho k , ta sẽ chọn ra một số tối thiểu các phần tử có tổng đồng dư với S theo modul k . Khi đó chỉ cần loại bỏ những phần tử này thì những phần tử còn lại sẽ là kết quả. Cách tiếp cận này cho phép tiết kiệm được không gian lưu trữ bởi số phần tử tối thiểu cần loại bỏ bao giờ cũng nhỏ hơn k .

Công thức truy hồi: Nếu ta gọi $f[i, t]$ là số phần tử tối thiểu phải chọn trong dãy $a[1..i]$ để có tổng chia k dư t . Nếu không có phương án chọn ta coi $f[i, t] = +\infty$. Khi đó $f[i, t]$ được tính qua công thức truy hồi sau:

- ❖ Nếu trong dãy trên không phải chọn $a[i]$ thì $f[i, t] = f[i - 1, t]$;
- ❖ Nếu trong dãy trên phải chọn $a[i]$ thì $f[i, t] = 1 + f[i - 1, \overline{t - A[i]}]$ ($\overline{t - A[i]}$ ở đây hiểu là phép trừ trên các lớp đồng dư mod k . Ví dụ khi $k = 7$ thì $\overline{1-3}=5$)

Từ trên suy ra $f[i, t] = \min(f[i - 1, t], 1 + f[i - 1, \overline{t - A[i]}])$

Cơ sở quy hoạch động: $f[0, 0] = 0$; $f[0, i] = +\infty$ (với $\forall i: 1 \leq i < k$).

P_3_03_5.PAS * Dãy con có tổng chia hết cho k

```
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program SubSequence;
const
  InputFile = 'SUBSEQ.INP';
  OutputFile = 'SUBSEQ.OUT';
  maxN = 1000;
  maxK = 1000;
var
  a: array[1..maxN] of Integer;
  f: array[0..maxN, 0..maxK - 1] of Integer;
  n, k: Integer;

procedure Enter;
var
  fi: Text;
  i: Integer;
```

```

begin
  Assign(fi, InputFile); Reset(fi);
  ReadLn(fi, n, k);
  for i := 1 to n do Read(fi, a[i]);
  Close(fi);
end;

function Sub(x, y: Integer): Integer; {Tính x - y (theo mod k)}
var
  tmp: Integer;
begin
  tmp := (x - y) mod k;
  if tmp >= 0 then Sub := tmp
  else Sub := tmp + k;
end;

procedure Optimize;
var
  i, t: Integer;
begin
  {Khởi tạo}
  f[0, 0] := 0;
  for t := 1 to k - 1 do f[0, t] := maxK;
  {Giải công thức truy hồi}
  for i := 1 to n do
    for t := 0 to k - 1 do {Tính f[i, t] := min (f[i - 1, t], f[i - 1, Sub(t, a[i])] + 1)}
      if f[i - 1, t] < f[i - 1, Sub(t, a[i])] + 1 then
        f[i, t] := f[i - 1, t]
      else
        f[i, t] := f[i - 1, Sub(t, a[i])] + 1;
end;

procedure Result;
var
  fo: Text;
  i, t: Integer;
  SumAll, Sum: Integer;
begin
  SumAll := 0;
  for i := 1 to n do SumAll := SumAll + a[i];
  Assign(fo, OutputFile); Rewrite(fo);
  WriteLn(fo, n - f[n, SumAll mod k]); {n - số phần tử bỏ đi = số phần tử giữ lại}
  i := n; t := SumAll mod k;
  Sum := 0;
  for i := n downto 1 do
    if f[i, t] = f[i - 1, t] then {Nếu phương án tối ưu không bỏ ai, tức là có chọn ai}
      begin
        WriteLn(fo, 'a[' , i, '] = ', a[i]);
        Sum := Sum + a[i];
      end
    else
      t := Sub(t, a[i]);
    WriteLn(fo, 'Sum = ', Sum);
    Close(fo);
end;

begin
  Enter;
  Optimize;
  Result;
end.

```

3.4.2. Cách giải 2

Phân các phần tử trong dãy A theo các lớp đồng dư modul k. Lớp i gồm các phần tử chia k dư i. Gọi Count[i] là số lượng các phần tử thuộc lớp i.

Với $0 \leq i, t < k$; Gọi $f[i, t]$ là số phần tử nhiều nhất có thể chọn được trong các lớp $0, 1, 2, \dots, i$ để được tổng chia k dư t. Trong trường hợp có cách chọn, gọi $\text{Trace}[i, t]$ là số phần tử được chọn trong lớp i theo phương án này, trong trường hợp không có cách chọn, $\text{Trace}[i, t]$ được coi là -1.

Ta dễ thấy rằng $f[0, 0] = \text{Count}[0]$, $\text{Trace}[0, 0] = \text{Count}[0]$, còn $\text{Trace}[0, i]$ với $i \neq 0$ bằng -1.

Với $i \geq 1; 0 \leq t < k$, Giả sử phương án chọn ra nhiều phần tử nhất trong các lớp từ 0 tới i để được tổng chia k dư t có lấy j phần tử của lớp i ($0 \leq j \leq \text{Count}[i]$), khi đó nếu bỏ j phần tử này đi, sẽ phải thu được phương án chọn ra nhiều phần tử nhất trong các lớp từ 0 tới $i - 1$ để được tổng chia k dư $t - i * j$. Từ đó suy ra công thức truy hồi:

$$f[i, t] = \max_{\substack{0 \leq j \leq \text{Count}[i] \\ \text{Trace}[i-1, t-j.i] \neq -1}} (f[i-1, t-j.i] + j)$$

$$\text{Trace}[i, t] = \arg \max_{\substack{0 \leq j \leq \text{Count}[i] \\ \text{Trace}[i-1, t-j.i] \neq -1}} (f[i-1, t-j.i] + j)$$

```
P_3_03_6.PAS * Dãy con có tổng chia hết cho k
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program SubSequence;
const
  InputFile = 'SUBSEQ.INP';
  OutputFile = 'SUBSEQ.OUT';
  maxN = 1000;
  maxK = 1000;
var
  a: array[1..maxN] of Integer;
  Count: array[0..maxK - 1] of Integer;
  f, Trace: array[0..maxK - 1, 0..maxK - 1] of Integer;
  n, k: Integer;

procedure Enter;
var
  fi: Text;
  i: Integer;
begin
  Assign(fi, InputFile); Reset(fi);
  ReadLn(fi, n, k);
  FillChar(Count, SizeOf(Count), 0);
  for i := 1 to n do
    begin
      Read(fi, a[i]);
      Inc(Count[a[i] mod k]); {Nhập dữ liệu đồng thời với việc tính các Count[.]}
    end;
  Close(fi);
end;

function Sub(x, y: Integer): Integer;
var
  tmp: Integer;
begin
  if x > y then
    Sub := y
  else
    Sub := x;
end;
```

```

tmp := (x - y) mod k;
if tmp >= 0 then Sub := tmp
else Sub := tmp + k;
end;

procedure Optimize;
var
  i, j, t: Integer;
begin
  FillChar(f, SizeOf(f), 0);
  f[0, 0] := Count[0];
  FillChar(Trace, SizeOf(Trace), $FF); {Khởi tạo các phần tử mảng Trace=-1}
  Trace[0, 0] := Count[0]; {Ngoại trừ Trace[0, 0] = Count[0]}
  for i := 1 to k - 1 do
    for t := 0 to k - 1 do
      for j := 0 to Count[i] do
        if (Trace[i - 1, Sub(t, j * i)] <> -1) and
           (f[i, t] < f[i - 1, Sub(t, j * i)] + j) then
          begin
            f[i, t] := f[i - 1, Sub(t, j * i)] + j;
            Trace[i, t] := j;
          end;
    end;
  end;

procedure Result;
var
  fo: Text;
  i, t, j: Integer;
  Sum: Integer;
begin
  t := 0;
  {Tính lại các Count[i] := Số phần tử phương án tối ưu sẽ chọn trong lớp i}
  for i := k - 1 downto 0 do
    begin
      j := Trace[i, t];
      t := Sub(t, j * i);
      Count[i] := j;
    end;
  Assign(fo, OutputFile); Rewrite(fo);
  WriteLn(fo, f[k - 1, 0]);
  Sum := 0;
  for i := 1 to n do
    begin
      t := a[i] mod k;
      if Count[t] > 0 then
        begin
          WriteLn(fo, 'a[' , i, '] = ', a[i]);
          Dec(Count[t]);
          Sum := Sum + a[i];
        end;
    end;
  WriteLn(fo, 'Sum = ', Sum);
  Close(fo);
end;

begin
  Enter;
  Optimize;
  Result;
end.

```

Cách giải thứ hai tốt hơn cách giải thứ nhất vì nó có thể thực hiện với n lớn. Ví dụ này cho thấy một bài toán quy hoạch động có thể có nhiều cách đặt công thức truy hồi để giải.

3.5. PHÉP NHÂN TỔ HỢP DÃY MA TRẬN

Với ma trận $A=\{a[i, j]\}$ kích thước $p \times q$ và ma trận $B=\{b[i, j]\}$ kích thước $q \times r$. Người ta có phép nhân hai ma trận đó để được ma trận $C=\{c[i, j]\}$ kích thước $p \times r$. Mỗi phần tử của ma trận C được tính theo công thức:

$$c[i, j] = \sum_{k=1}^q a[i, j] \cdot b[k, j], (1 \leq i \leq p; 1 \leq j \leq r)$$

Ví dụ:

A là ma trận kích thước 3×4 , B là ma trận kích thước 4×5 thì C sẽ là ma trận kích thước 3×5

$$\left(\begin{array}{cccc} 1 & 2 & 3 & 4 \\ 5 & 6 & 7 & 8 \\ 9 & 10 & 11 & 12 \end{array} \right) \times \left(\begin{array}{ccc|c} 1 & 0 & 2 & 4 \\ 0 & 1 & 0 & 5 \\ 3 & 0 & 1 & 6 \\ \hline 1 & 1 & 1 & 1 \end{array} \right) = \left(\begin{array}{ccc|c|c} 14 & 6 & 9 & 36 & 9 \\ 34 & 14 & 25 & 100 & 21 \\ 54 & 22 & 41 & 164 & 33 \end{array} \right)$$

Để thực hiện phép nhân hai ma trận $A(p \times q)$ và $B(q \times r)$ ta có thể làm như đoạn chương trình sau:

```
for i := 1 to p do
  for j := 1 to r do
    begin
      c[i, j] := 0;
      for k := 1 to q do c[i, j] := c[i, j] + a[i, k] * b[k, j];
    end;
```

Phí tổn để thực hiện phép nhân ma trận có thể đánh giá qua số lần thực hiện phép nhân số học, với giải thuật nhân hai ma trận kể trên, để nhân ma trận A cấp $p \times q$ với ma trận B cấp $q \times r$ ta cần thực hiện $p \cdot q \cdot r$ phép nhân số học.

Phép nhân ma trận không có tính chất giao hoán nhưng có tính chất kết hợp

$$(A \cdot B) \cdot C = A \cdot (B \cdot C)$$

Vậy nếu A là ma trận cấp 3×4 , B là ma trận cấp 4×10 và C là ma trận cấp 10×15 thì:

- ❖ Để tính $(A \cdot B) \cdot C$, phép tính $(A \cdot B)$ cho ma trận kích thước 3×10 sau $3 \cdot 4 \cdot 10 = 120$ phép nhân số, sau đó nhân tiếp với C được ma trận kết quả kích thước 3×15 sau $3 \cdot 10 \cdot 15 = 450$ phép nhân số. Vậy tổng số phép nhân số học phải thực hiện sẽ là 570.
- ❖ Để tính $A \cdot (B \cdot C)$, phép tính $(B \cdot C)$ cho ma trận kích thước 4×15 sau $4 \cdot 10 \cdot 15 = 600$ phép nhân số, lấy A nhân với ma trận này được ma trận kết quả kích thước 3×15 sau $3 \cdot 4 \cdot 15 = 180$ phép nhân số. Vậy tổng số phép nhân số học phải thực hiện sẽ là 780.

Vậy thì trình tự thực hiện có ảnh hưởng lớn tới chi phí. Vấn đề đặt ra là tính số phí tổn ít nhất

khi thực hiện phép nhân một dãy các ma trận: $\prod_{i=1}^n m[i] = m[1] \cdot m[2] \dots \cdot m[n]$

Với :

$m[1]$ là ma trận kích thước $a[1] \times a[2]$

$m[2]$ là ma trận kích thước $a[2] \times a[3]$

...

$m[n]$ là ma trận kích thước $a[n] \times a[n+1]$

Input: file văn bản MULTMAT.INP

- ❖ Dòng 1: Chứa số nguyên dương $n \leq 100$
- ❖ Dòng 2: Chứa $n + 1$ số nguyên dương $a[1], a[2], \dots, a[n+1]$ ($\forall i: 1 \leq a[i] \leq 100$) cách nhau ít nhất một dấu cách

Output: file văn bản MULTMAT.OUT

- ❖ Dòng 1: Ghi số phép nhân số học tối thiểu cần thực hiện
- ❖ Dòng 2: Ghi biểu thức kết hợp tối ưu của phép nhân dãy ma trận

MULTMAT.INP	MULTMAT.OUT
6	Number of numerical multiplications: 31
3 2 3 1 2 2 3	((m[1].(m[2].m[3])).((m[4].m[5]).m[6]))

Trước hết, nếu dãy chỉ có một ma trận thì chi phí bằng 0, tiếp theo ta nhận thấy chi phí để nhân một cặp ma trận có thể tính được ngay. Vậy có thể ghi nhận được chi phí cho phép nhân hai ma trận liên tiếp bất kỳ trong dãy. Sử dụng những thông tin đã ghi nhận để tối ưu hoá phí tổn nhân những bộ ba ma trận liên tiếp ... Cứ tiếp tục như vậy cho tới khi ta tính được phí tổn nhân n ma trận liên tiếp.

3.5.1. Công thức truy hồi:

Gọi $f[i, j]$ là số phép nhân số học tối thiểu cần thực hiện để nhân đoạn ma trận liên tiếp:

$$\prod_{t=i}^j m[t] = m[i].m[i+1] \dots .m[j]. \text{ Thì khi đó } f[i, i] = 0 \text{ với } \forall i.$$

Để tính $\prod_{t=i}^j m[t]$, có thể có nhiều cách kết hợp:

$$\prod_{t=i}^j m[t] = \left(\prod_{u=i}^k m[u] \right) \cdot \left(\prod_{v=k+1}^j m[v] \right); \forall k: i \leq k < j$$

Với một cách kết hợp (phụ thuộc vào cách chọn vị trí k), chi phí tối thiểu phải thực hiện bằng:

$f[i, k]$ (là chi phí tối thiểu tính $\prod_{u=i}^k m[u]$) cộng với $f[k+1, j]$ (là chi phí tối thiểu tính $\prod_{v=k+1}^j m[v]$) cộng với $a[j].a[k+1].a[j+1]$ (là chi phí thực hiện phép nhân cuối cùng giữa ma trận $\prod_{u=i}^k m[u]$ và ma trận $\prod_{v=k+1}^j m[v]$). Từ đó suy ra: do có nhiều cách kết hợp, mà ta cần chọn cách kết hợp

để có chi phí ít nhất nên ta sẽ cực tiểu hóa $f[i, j]$ theo công thức:

$$f[i, j] = \min_{1 \leq k < j} (f[i, k] + f[k+1, j] + a[i].a[k+1].a[j+1])$$

3.5.2. Tính bảng phương án

Bảng phương án F là bảng hai chiều, nhìn vào công thức truy hồi, ta thấy $f[i, j]$ chỉ được tính khi mà $f[i, k]$ cũng như $f[k+1, j]$ đều đã biết ($\forall k: i \leq k < j$). Tức là ban đầu ta điền cơ sở quy

hoạch động vào đường chéo chính của bảng($\forall i: f[i, i] := 0$), từ đó tính các giá trị thuộc đường chéo nằm phía trên (tính các $f[i, i + 1]$), rồi lại tính các giá trị thuộc đường chéo nằm phía trên nữa (các $f[i, i + 2]$) ... Đến khi tính được $f[1, n]$ thì dừng lại

3.5.3. Tìm cách kết hợp tối ưu

Tại mỗi bước tính $f[i, j]$, ta ghi nhận lại $Tr[i, j]$ là điểm k mà cách tính:

$$\prod_{t=i}^j m[t] = \left(\prod_{u=i}^k m[u] \right) \cdot \left(\prod_{v=k+1}^j m[v] \right)$$

cần số phép nhân số học ít nhất trên tất cả các cách chọn k. Sau đó, muốn in ra phép kết hợp tối ưu để nhân $\prod_{t=i}^j m[t]$, ta sẽ in ra cách kết hợp tối ưu để nhân $\prod_{q=i}^{Tr[i,j]} m[q]$ và cách kết hợp tối ưu để nhân $\prod_{r=Tr[i,j]+1}^j m[r]$ (có kèm theo dấu đóng mở ngoặc) đồng thời viết thêm dấu “.” vào giữa hai biểu thức đó.

```
P_3_03_7.PAS * Nhân tối ưu dây ma trận
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program MatrixMultiplications;
const
  InputFile = 'MULTMAT.INP';
  OutputFile = 'MULTMAT.OUT';
  max = 100;
var
  a: array[1..max + 1] of Integer;
  f: array[1..max, 1..max] of Integer;
  tr: array[1..max, 1..max] of Integer;
  n: Integer;
  fo: Text;

procedure Enter; {Nhập dữ liệu}
var
  i: Integer;
  fi: Text;
begin
  Assign(fi, InputFile); Reset(fi);
  ReadLn(fi, n);
  for i := 1 to n + 1 do Read(fi, a[i]);
  Close(fi);
end;

procedure Optimize; {Quy hoạch động}
var
  i, j, k, len: Integer;
  x, p, q, r: Integer;
begin
  {Điền cơ sở quy hoạch động vào bảng phương án}
  for i := 1 to n do
    for j := i to n do
      if i = j then f[i, j] := 0
      else f[i, j] := High(Integer);
  {Giải công thức truy hồi}
  for len := 2 to n do {Thù với các độ dài đoạn từ 2 tới n}
    for i := 1 to n - len + 1 do {Tính các f[i, i + len - 1]}
      begin
```

```

j := i + len - 1;
for k := i to j - 1 do {Thứ các vị trí phân hoạch k}
begin
  p := a[i]; q := a[k + 1]; r := a[j + 1];
  x := f[i, k] + f[k + 1, j] + p * q * r;
  if x < f[i, j] then {Tối ưu hoá f[i, j]}
    begin
      f[i, j] := x;
      tr[i, j] := k;
    end;
  end;
end;

procedure Trace(i, j: Integer); {Truy vết bằng đệ quy, thủ tục này in ra cách kết hợp tối ưu tính m[i]...m[j]}
var
  k: Integer;
begin
  if i = j then Write(fo, 'm['', i, ']')
  else
    begin
      Write(fo, '('');
      k := tr[i, j];
      Trace(i, k);
      Write(fo, '.');
      Trace(k + 1, j);
      Write(fo, ')');
    end;
end;

begin
  Enter;
  Optimize;
  Assign(fo, OutputFile); Rewrite(fo);
  WriteLn(fo, 'Number of numerical multiplications: ', f[1, n]);
  Trace(1, n);
  Close(fo);
end.

```

3.6. BÀI TẬP LUYỆN TẬP

3.6.1. Bài tập có hướng dẫn lời giải

Bài 1

Nhập vào hai số nguyên dương n và k ($n, k \leq 100$). Hãy cho biết

- Có bao nhiêu số nguyên dương có $\leq n$ chữ số mà tổng các chữ số đúng bằng k. Nếu có hơn 1 tỉ số thì chỉ cần thông báo có nhiều hơn 1 tỉ.
- Nhập vào một số $p \leq 1$ tỉ. Cho biết nếu đếm các số tìm được xếp theo thứ tự tăng dần thì số thứ p là số nào ?

Hướng dẫn:

Câu a: Ta sẽ đếm số các số có đúng n chữ số mà tổng các chữ số (TCCS) bằng k, chỉ có điều các số của ta cho phép có thể bắt đầu bằng 0. Ví dụ: ta coi 0045 là số có 4 chữ số mà TCCS là 9. Gọi F[n, k] là số các số có n chữ số mà TCCS bằng k. Các số đó có thể biểu diễn bằng mảng x[1..n] gồm các chữ số 0...9 và $x[1] + x[2] + \dots + x[n] = k$. Nếu cố định $x[1] = t$ thì ta nhận thấy $x[2..n]$ lập thành một số có $n - 1$ chữ số mà TCCS bằng $k - t$. Suy ra do $x[1]$ có thể

nhận các giá trị từ 0 tới 9 nên về mặt số lượng: $F[n, k] = \sum_{t=0}^9 F[n-1, k-t]$. Đây là công thức truy hồi tính $F[n, k]$, thực ra chỉ xét những giá trị t từ 0 tới 9 và $t \leq k$ mà thôi (để tránh trường hợp $k - t < 0$). Chú ý rằng nếu tại một bước nào đó tính ra một phần tử của $F > 10^9$ thì ta đặt lại phần tử đó là $10^9 + 1$ để tránh bị tràn số do cộng hai số quá lớn. Kết thúc quá trình tính toán, nếu $F[n, k] = 10^9 + 1$ thì ta chỉ cần thông báo chung chung là có > 1 tỉ số.

Cơ sở quy hoạch động thì có thể đặt là:

$F[1, k] =$ số các số có 1 chữ số mà TCCS bằng k , như vậy:

$$F[1, k] = \begin{cases} 1, & \text{if } 0 \leq k \leq 9 \\ 0, & \text{otherwise} \end{cases}$$

Câu b: Dựa vào bảng phương án $F[0..n, 0..k]$ để dò ra số mang thứ tự đã cho.

Bài 2

Cho n gói kẹo ($n \leq 200$), mỗi gói chứa không quá 200 viên kẹo, và một số $M \leq 40000$. Hãy chỉ ra một cách lấy ra một số các gói kẹo để được tổng số kẹo là M , hoặc thông báo rằng không thể thực hiện được việc đó.

Hướng dẫn:

Giả sử số kẹo chứa trong gói thứ i là $A[i]$

Gọi $b[V]$ là số nguyên dương bé nhất thoả mãn: Có thể chọn trong số các gói kẹo từ gói 1 đến gói $b[V]$ ra một số gói để được tổng số kẹo là V . Nếu không có phương án chọn, ta coi $b[V] = +\infty$. Trước tiên, khởi tạo $b[0] := 0$ và các $b[V] := +\infty$ với mọi $V > 0$.

Với một giá trị V , gọi k là giá trị cần tìm để gán cho $b[V]$, vì k cần bé nhất có thể, nên nếu có cách chọn trong số các gói kẹo từ gói 1 đến gói k để được số kẹo V thì chắc chắn phải chọn gói k . Khi đã chọn gói k rồi thì trong số các gói kẹo từ 1 đến $k - 1$, phải chọn ra được một số gói để được số kẹo là $V - A[k]$. Tức là $b[V - A[k]] \leq k - 1 < k$.

Suy ra $b[V]$ sẽ được tính bằng cách:

Xét tất cả các gói kẹo k có $A[k] \leq V$ và thoả mãn $b[V - A[k]] < k$, chọn ra chỉ số k bé nhất gán cho $b[V]$. Đây chính là công thức truy hồi tính bảng phương án.

$$b[V] = \min \left\{ k \mid (A[k] \leq V) \wedge (b[V - A[k]] < k) \right\}$$

Sau khi đã tính hết dãy $b[1..M]$. Nếu $b[M]$ vẫn bằng $+\infty$ thì có nghĩa là không có phương án chọn. Nếu không thì sẽ chọn gói $p[1] = b[M]$, tiếp theo sẽ chọn gói $p[2] = b[M - A[p[1]]]$, rồi lại chọn gói $p[3] = b[M - A[p[1]] - A[p[2]]]$... Đến khi truy vết về tới $b[0]$ thì thôi.

Bài 3

Cho n gói kẹo ($n \leq 200$), mỗi gói chứa không quá 200 viên kẹo, hãy chia các gói kẹo ra làm hai nhóm sao cho số kẹo giữa hai nhóm chênh lệch nhau ít nhất

Hướng dẫn:

Gọi S là tổng số kẹo và M là nửa tổng số kẹo, áp dụng cách giải như bài 2. Sau đó

Tìm số nguyên dương T thoả mãn:

- ❖ $T \leq M$
- ❖ Tồn tại một cách chọn ra một số gói kẹo để được tổng số kẹo là T ($b[T] \neq +\infty$)
- ❖ T lớn nhất có thể

Sau đó chọn ra một số gói kẹo để được T viên kẹo, các gói kẹo đó được đưa vào một nhóm, số còn lại vào nhóm thứ hai.

Bài 4

Cho một bảng A kích thước $m \times n$, trên đó ghi các số nguyên. Một người xuất phát tại ô nào đó của cột 1, cần sang cột n (tại ô nào cũng được). Quy tắc: Từ ô $A[i, j]$ chỉ được quyền sang một trong 3 ô $A[i, j + 1]$; $A[i - 1, j + 1]$; $A[i + 1, j + 1]$. Hãy tìm vị trí ô xuất phát và hành trình đi từ cột 1 sang cột n sao cho tổng các số ghi trên đường đi là lớn nhất.

	1	2	6	7	9
	7	6	5	6	7
	1	2	3	4	2
	4	7	8	7	6

Hướng dẫn:

Gọi $B[i, j]$ là số điểm lớn nhất có thể có được khi tới ô $A[i, j]$. Rõ ràng đối với những ô ở cột 1 thì $B[i, 1] = A[i, 1]$:

1	2	6	7	9
7	6	5	6	7
1	2	3	4	2
4	7	8	7	6

1				
7				
1				
4				

Với những ô (i, j) ở các cột khác. Vì chỉ những ô $(i, j - 1)$, $(i - 1, j - 1)$, $(i + 1, j - 1)$ là có thể sang được ô (i, j) , và khi sang ô (i, j) thì số điểm được cộng thêm $A[i, j]$ nữa. Chúng ta cần $B[i, j]$ là số điểm lớn nhất có thể nên $B[i, j] = \max(B[i, j - 1], B[i - 1, j - 1], B[i + 1, j - 1]) + A[i, j]$. Ta dùng công thức truy hồi này tính tất cả các $B[i, j]$. Cuối cùng chọn ra $B[i, n]$ là phần tử lớn nhất trên cột n của bảng B và từ đó truy vết ra đường đi nhiều điểm nhất.

3.6.2. Bài tập tự làm

Bài 1

Lập trình giải bài toán cái túi với kích thước dữ liệu: $n \leq 10000$; $M \leq 10000$ và giới hạn bộ nhớ 10MB.

Bài 2

Xâu ký tự S gọi là xâu con của xâu ký tự T nếu có thể xoá bớt một số ký tự trong xâu T để được xâu S. Lập chương trình nhập vào hai xâu ký tự X, Y. Tìm xâu Z có độ dài lớn nhất là xâu con của cả X và Y. Ví dụ: X = 'abcdefghijklm'; Y = 'abc1def2ghi3' thì Z là 'abcdefg'.

Bài 3

Một xâu ký tự X gọi là **chứa** xâu ký tự Y nếu như có thể xoá bớt một số ký tự trong xâu X để được xâu Y: Ví dụ: Xâu '1a2b3c45d' chứa xâu '12345'. Một xâu ký tự gọi là **đôi** xứng nếu nó không thay đổi khi ta viết các ký tự trong xâu theo thứ tự ngược lại: Ví dụ: 'abcABADABAcb', 'MADAM' là các xâu **đôi** xứng.

Nhập một xâu ký tự S có độ dài không quá 128, hãy tìm xâu ký tự T thoả mãn cả 3 điều kiện:

- ❖ Đôi xứng
- ❖ Chứa xâu S
- ❖ Có ít ký tự nhất (có độ dài ngắn nhất)

Nếu có nhiều xâu T thoả mãn đồng thời 3 điều kiện trên thì chỉ cần cho biết một. Chẳng hạn với S = 'a_101_b' thì chọn T = 'ab_101_ba' hay T = 'ba_101_ab' đều đúng.

Ví dụ:


S	T
MADAM	MADAM
Edbabed	edcbabcde
00_11_22_33_222_1_000	000_11_222_33_222_11_000
abcdefg_hh_gfe_1_d_2_c_3_ba	ab_3_c_2_d_1_efg_hh_gfe_1_d_2_c_3_ba

Bài 4

Có n loại tiền giấy: Tờ giấy bạc loại i có mệnh giá là $V[i]$ ($n \leq 20, 1 \leq V[i] \leq 10000$). Hỏi muôn mua một món hàng giá là M thì có bao nhiêu cách trả số tiền đó bằng những loại giấy bạc đã cho (Trường hợp có > 1 tỉ cách thì chỉ cần thông báo có nhiều hơn 1 tỉ). Nếu tồn tại cách trả, cho biết cách trả phải dùng ít tờ tiền nhất.

Bài 5

Cho n quân đô-mi-nô xếp dựng đứng theo hàng ngang và được đánh số từ 1 đến n. Quân đô-mi-nô thứ i có số ghi ở ô trên là $a[i]$ và số ghi ở ô dưới là $b[i]$. Xem hình vẽ:



Biết rằng $1 \leq n \leq 100$ và $0 \leq a[i], b[i] \leq 6$ với $\forall i: 1 \leq i \leq n$. Cho phép lật ngược các quân đô-mi-nô. Khi một quân đô-mi-nô thứ i bị lật, nó sẽ có số ghi ở ô trên là $b[i]$ và số ghi ở ô dưới là $a[i]$.

Vấn đề đặt ra là hãy tìm cách lật các quân đô-mi-nô sao cho chênh lệch giữa tổng các số ghi ở hàng trên và tổng các số ghi ở hàng dưới là tối thiểu. Nếu có nhiều phương án lật tốt nhau, thì chỉ ra phương án phải lật ít quân nhất.

Như ví dụ trên thì sẽ lật hai quân Đô-mi-nô thứ 5 và thứ 6. Khi đó:

$$\text{Tổng các số ở hàng trên} = 1 + 1 + 4 + 4 + 6 + 1 = 17$$

Tổng các số ở hàng dưới = $6 + 3 + 1 + 1 + 0 + 6 = 17$

Bài 6

Xét bảng H kích thước 4×4 , các hàng và các cột được đánh chỉ số A, B, C, D. Trên 16 ô của bảng, mỗi ô ghi 1 ký tự A hoặc B hoặc C hoặc D.

	A	B	C	D
A	A	A	B	B
B	C	D	A	B
C	B	C	B	A
D	B	D	D	D

Cho xâu S gồm n ký tự chỉ gồm các chữ A, B, C, D.

Xét phép co R(i): thay ký tự $S[i]$ và $S[i+1]$ bởi ký tự nằm trên hàng $S[i]$, cột $S[i+1]$ của bảng H.

Ví dụ: $S = ABCD$; áp dụng liên tiếp 3 lần $R(1)$ sẽ được

$$ABCD \rightarrow ACD \rightarrow BD \rightarrow B.$$

Yêu cầu: Cho trước một ký tự $X \in \{A, B, C, D\}$, hãy chỉ ra thứ tự thực hiện $n - 1$ phép co để ký tự còn lại cuối cùng trong S là X.

Bài 7

Cho N số tự nhiên $a[1], a[2], \dots, a[n]$. Biết rằng $1 \leq n \leq 200$ và $0 \leq a[i] \leq 200$. Ban đầu các số được đặt liên tiếp theo đúng thứ tự cách nhau bởi dấu “?”: $a[1] ? a[2] ? \dots ? a[n]$. Yêu cầu: Cho trước số nguyên K, hãy tìm cách thay các dấu “?” bằng dấu cộng hay dấu trừ để được một biểu thức số học cho giá trị là K. Biết rằng $1 \leq n \leq 200$ và $0 \leq a[i] \leq 100$.

Ví dụ: Ban đầu $1 ? 2 ? 3 ? 4$ và $K = 0$ sẽ cho kết quả $1 - 2 - 3 + 4$.

Bài 8

Dãy Catalan là một dãy số tự nhiên bắt đầu là 0, kết thúc là 0, hai phần tử liên tiếp hơn kém nhau 1 đơn vị. Hãy lập chương trình nhập vào số nguyên dương n lẻ và một số nguyên dương p. Cho biết rằng nếu như ta đếm tất cả các dãy Catalan độ dài n xếp theo thứ tự từ điển thì dãy thứ p là dãy nào.

Một bài toán quy hoạch động có thể có nhiều cách tiếp cận khác nhau, chọn cách nào là tuỳ theo yêu cầu bài toán sao cho dễ dàng cài đặt nhất. Phương pháp này thường không khó khăn trong việc tính bảng phương án, không khó khăn trong việc tìm cơ sở quy hoạch động, mà khó khăn chính là **nhìn nhận ra bài toán quy hoạch động và tìm ra công thức truy hồi** giải nó, công việc này đòi hỏi sự nhanh nhẹn, khôn khéo, mà chỉ từ sự rèn luyện mới có thể có được. Hãy đọc lại §1 để tìm hiểu kỹ các phương pháp thông dụng khi cài đặt một chương trình giải công thức truy hồi.



PHẦN 4. CÁC THUẬT TOÁN TRÊN ĐỒ THỊ



Leonhard Euler
(1707-1783)

Trên thực tế có nhiều bài toán liên quan tới một tập các đối tượng và những mối liên hệ giữa chúng, đòi hỏi toán học phải đặt ra một mô hình biểu diễn một cách chặt chẽ và tổng quát bằng ngôn ngữ ký hiệu, đó là đồ thị. Những ý tưởng cơ bản của nó được đưa ra từ thế kỷ thứ XVIII bởi nhà toán học Thụy Sĩ Leonhard Euler, ông đã dùng mô hình đồ thị để giải bài toán về những cây cầu Konigsberg nổi tiếng.

Mặc dù Lý thuyết đồ thị đã được khoa học phát triển từ rất lâu nhưng lại có nhiều ứng dụng hiện đại. Đặc biệt trong khoảng vài mươi năm trở lại đây, cùng với sự ra đời của máy tính điện tử và sự phát triển nhanh chóng của Tin học, Lý thuyết đồ thị càng được quan tâm đến nhiều hơn. Đặc biệt là các thuật toán trên đồ thị đã có nhiều ứng dụng trong nhiều lĩnh vực khác nhau như: Mạng máy tính, Lý thuyết mã, Tối ưu hoá, Kinh tế học v.v... Hiện nay, môn học này là một trong những kiến thức cơ sở của bộ môn khoa học máy tính.

Trong phạm vi một chuyên đề, không thể nói kỹ và nói hết những vấn đề của lý thuyết đồ thị. Tập bài giảng này sẽ xem xét lý thuyết đồ thị dưới góc độ người lập trình, tức là khảo sát những **thuật toán cơ bản nhất** có thể **dễ dàng cài đặt trên máy tính** một số ứng dụng của nó.. Công việc của người lập trình là đọc hiểu được ý tưởng cơ bản của thuật toán và cài đặt được chương trình trong bài toán tổng quát cũng như trong trường hợp cụ thể.

§1. CÁC KHÁI NIỆM CƠ BẢN


1.1. ĐỊNH NGHĨA ĐỒ THỊ (GRAPH)

Là một cấu trúc rời rạc gồm các đỉnh và các cạnh nối các đỉnh đó. Được mô tả hình thức:

$$G = (V, E)$$

V gọi là tập các **đỉnh** (Vertices) và E gọi là tập các **cạnh** (Edges). Có thể coi E là tập các cặp (u, v) với u và v là hai đỉnh của V.

Một số hình ảnh của đồ thị:



Hình 52: Ví dụ về mô hình đồ thị

Có thể phân loại đồ thi theo đặc tính và số lượng của tập các cạnh E:

Cho đồ thị $G = (V, E)$. Định nghĩa một cách hình thức


G được gọi là **đơn đồ thị** nếu giữa hai đỉnh u, v của V có nhiều nhất là 1 cạnh trong E nối từ u tới v.

G được gọi là **đa đồ thị** nếu giữa hai đỉnh u, v của V có thể có nhiều hơn 1 cạnh trong E nối từ u tới v (Hiển nhiên đơn đồ thi cũng là đa đồ thi).

G được gọi là đồ thị **vô hướng** (undirected graph) nếu các cạnh trong E là không định hướng, tức là cạnh nối hai đỉnh u, v bất kỳ cũng là cạnh nối hai đỉnh v, u . Hay nói cách khác, tập E gồm các cặp (u, v) không tính thứ tự. $(u, v) \equiv (v, u)$

G được gọi là đồ thị **có hướng** (directed graph) nếu các cạnh trong E là có định hướng, có thể có cạnh nối từ đỉnh u tới đỉnh v nhưng chưa chắc đã có cạnh nối từ đỉnh v tới đỉnh u . Hay nói cách khác, tập E gồm các cặp (u, v) có tính thứ tự: $(u, v) \neq (v, u)$. Trong đồ thị có hướng, các cạnh được gọi là các **cung**. Đồ thị vô hướng cũng có thể coi là đồ thị có hướng nếu ta coi cạnh nối hai đỉnh u, v bất kỳ tương đương với hai cung (u, v) và (v, u) .

Ví dụ:

**Hình 53: Phân loại đồ thị**

1.2. CÁC KHÁI NIỆM

Như trên định nghĩa **đồ thị** $G = (V, E)$ là một cấu trúc rời rạc, tức là các tập V và E hoặc là tập hữu hạn, hoặc là tập đếm được, có nghĩa là ta có thể đánh số thứ tự 1, 2, 3... cho các phần tử của tập V và E . Hơn nữa, đứng trên phương diện người lập trình cho máy tính thì ta chỉ quan tâm đến các đồ thị hữu hạn (V và E là tập hữu hạn) mà thôi, chính vì vậy từ đây về sau, nếu không chú thích gì thêm thì khi nói tới đồ thị, ta hiểu rằng đó là đồ thị hữu hạn.

1.2.1. Cạnh liên thuộc, đỉnh kề, bậc

Đối với đồ thị vô hướng $G = (V, E)$. Xét một cạnh $e \in E$, nếu $e = (u, v)$ thì ta nói hai đỉnh u và v là **kề nhau (adjacent)** và cạnh e này **liên thuộc (incident)** với đỉnh u và đỉnh v .

Với một đỉnh v trong đồ thị, ta định nghĩa **bậc (degree)** của v , ký hiệu $\deg(v)$ là số cạnh liên thuộc với v . Dễ thấy rằng trên đơn đồ thị thì số cạnh liên thuộc với v cũng là số đỉnh kề với v .

Định lý: Giả sử $G = (V, E)$ là đồ thị vô hướng với m cạnh, khi đó tổng tất cả các bậc đỉnh trong V sẽ bằng $2m$:

$$\sum_{v \in V} \deg(v) = 2m$$

Chứng minh: Khi lấy tổng tất cả các bậc đỉnh tức là mỗi cạnh $e = (u, v)$ bất kỳ sẽ được tính một lần trong $\deg(u)$ và một lần trong $\deg(v)$. Từ đó suy ra kết quả.

Hệ quả: Trong đồ thị vô hướng, số đỉnh bậc lẻ là số chẵn

Đối với đồ thị có hướng $G = (V, E)$. Xét một cung $e \in E$, nếu $e = (u, v)$ thì ta nói **u nối tới v** và **v nối từ u**, cung e là **đi ra khỏi đỉnh u và đi vào đỉnh v**. Đỉnh u khi đó được gọi là **đỉnh đầu**, đỉnh v được gọi là **đỉnh cuối** của cung e .

Với mỗi đỉnh v trong đồ thị có hướng, ta định nghĩa: **Bán bậc ra (out-degree)** của v ký hiệu $\deg^+(v)$ là số cung đi ra khỏi nó; **bán bậc vào (in-degree)** ký hiệu $\deg^-(v)$ là số cung đi vào đỉnh đó

Định lý: Giả sử $G = (V, E)$ là đồ thị có hướng với m cung, khi đó tổng tất cả các bán bậc ra của các đỉnh bằng tổng tất cả các bán bậc vào và bằng m :

$$\sum_{v \in V} \deg^+(v) = \sum_{v \in V} \deg^-(v) = m$$

Chứng minh: Khi lấy tổng tất cả các bán bậc ra hay bán bậc vào, mỗi cung (u, v) bất kỳ sẽ được tính đúng 1 lần trong $\deg^+(u)$ và cũng được tính đúng 1 lần trong $\deg^-(v)$. Từ đó suy ra kết quả

1.2.2. Đường đi và chu trình

Một đường đi với độ dài p là một dãy $P=\langle v_0, v_1, \dots, v_p \rangle$ của các đỉnh sao cho $(v_{i-1}, v_i) \in E$, ($\forall i: 1 \leq i \leq p$). Ta nói đường đi P **bao gồm** các đỉnh v_0, v_1, \dots, v_p và các cạnh $(v_0, v_1), (v_1, v_2), \dots, (v_{p-1}, v_p)$. Nếu có một đường đi như trên thì ta nói v_p **đến được (reachable)** từ v_0 qua P . Một đường đi gọi là **đơn giản (simple)** nếu tất cả các đỉnh trên đường đi là hoàn toàn phân biệt, một **đường đi con (subpath)** P' của P là một đoạn liên tục của các dãy các đỉnh dọc theo P . Đường đi P trở thành **chu trình (circuit)** nếu $v_0=v_p$. Chu trình P gọi là **đơn giản (simple)** nếu v_1, v_2, \dots, v_p là hoàn toàn phân biệt

1.2.3. Một số khái niệm khác

Hai đồ thị $G = (V, E)$ và $G' = (V', E')$ được gọi là **đẳng cấu (isomorphic)** nếu tồn tại một song ánh $f: V \rightarrow V'$ sao cho $(u, v) \in E$ nếu và chỉ nếu $(f(u), f(v)) \in E'$.

Đồ thị $G' = (V', E')$ là **đồ thị con (subgraph)** của đồ thị $G = (V, E)$ nếu $V' \subseteq V$ và $E' \subseteq E$. Khi đó G' được gọi là **đồ thị con cảm ứng (induced)** từ G bởi V' nếu $E' = \{(u, v) \in E | u, v \in V'\}$

Cho một đồ thị vô hướng $G = (V, E)$, ta gọi **phiên bản có hướng (directed version)** của G là một đồ thị có hướng $G' = (V, E')$ sao cho $(u, v) \in E'$ nếu và chỉ nếu $(u, v) \in E$. Nói cách khác G' được tạo thành từ G bằng cách thay mỗi cạnh bằng hai cung có hướng ngược chiều nhau.

Cho một đồ thị có hướng $G = (V, E)$, ta gọi **phiên bản vô hướng (undirected version)** của G là một đồ thị vô hướng $G' = (V, E')$ sao cho $(u, v) \in E'$ nếu và chỉ nếu $(u, v) \in E$ hoặc $(v, u) \in E$.

Một đồ thị vô hướng gọi là **liên thông (connected)** nếu với mọi cặp đỉnh (u, v) ta có u đến được v . Một đồ thị có hướng gọi là **liên thông mạnh (strongly connected)** nếu với mỗi cặp đỉnh (u, v) , ta có u đến được v và v đến được u . Một đồ thị có hướng gọi là **liên thông yếu (weakly connected)** nếu phiên bản vô hướng của nó là đồ thị liên thông.

Một đồ thị vô hướng được gọi là **đầy đủ (complete)** nếu mọi cặp đỉnh đều là kề nhau. Một đồ thị vô hướng gọi là **hai phia (bipartite)** nếu tập đỉnh của nó có thể chia làm hai tập rời nhau X, Y sao cho không tồn tại cạnh nối hai đỉnh $\in X$ cũng như không tồn tại cạnh nối hai đỉnh $\in Y$.

Người ta còn mở rộng khái niệm đồ thị thành **siêu đồ thị (hypergraph)**, một siêu đồ thị tương tự như đồ thị vô hướng, những mỗi **siêu cạnh (hyperedge)** không những chỉ có thể nối hai đỉnh mà còn có thể nối một tập các đỉnh với nhau.

§2. BIỂU DIỄN ĐỒ THỊ TRÊN MÁY TÍNH

2.1. MA TRẬN KÈ (ADJACENCY MATRIX)


Giả sử $G = (V, E)$ là một **đơn đồ thị** có số đỉnh (ký hiệu $|V|$) là n , Không mất tính tổng quát có thể coi các đỉnh được đánh số $1, 2, \dots, n$. Khi đó ta có thể biểu diễn đồ thị bằng một ma trận vuông $A = [a[i, j]]$ cấp n . Trong đó:

- ❖ $a[i, j] = 1$ nếu $(i, j) \in E$
- ❖ $a[i, j] = 0$ nếu $(i, j) \notin E$

Với $\forall i$, giá trị của $a[i, i]$ có thể đặt tùy theo mục đích, thông thường nên đặt bằng 0;

Đối với đa đồ thị thì việc biểu diễn cũng tương tự trên, chỉ có điều nếu như (i, j) là cạnh thì không phải ta ghi số 1 vào vị trí $a[i, j]$ mà là ghi số cạnh nối giữa đỉnh i và đỉnh j .

Ví dụ:



Các tính chất của ma trận kè:

Đối với đồ thị vô hướng G , thì ma trận kè tương ứng là ma trận đối xứng ($a[i, j] = a[j, i]$), điều này không đúng với đồ thị có hướng.

Nếu G là đồ thị vô hướng và A là ma trận kè tương ứng thì trên ma trận A :

$$\text{Tổng các số trên hàng } i = \text{Tổng các số trên cột } i = \text{Bậc của đỉnh } i = \deg(i)$$

Nếu G là đồ thị có hướng và A là ma trận kè tương ứng thì trên ma trận A :

$$\text{Tổng các số trên hàng } i = \text{Bán bậc ra của đỉnh } i = \deg^+(i)$$

$$\text{Tổng các số trên cột } i = \text{Bán bậc vào của đỉnh } i = \deg^-(i)$$

Trong trường hợp G là đơn đồ thị, ta có thể biểu diễn ma trận kè A tương ứng là các phần tử logic. $a[i, j] = \text{TRUE}$ nếu $(i, j) \in E$ và $a[i, j] = \text{FALSE}$ nếu $(i, j) \notin E$

Ưu điểm của ma trận kè:


- ❖ Đơn giản, trực quan, dễ cài đặt trên máy tính
- ❖ Để kiểm tra xem hai đỉnh (u, v) của đồ thị có kè nhau hay không, ta chỉ việc kiểm tra bằng một phép so sánh: $a[u, v] \neq 0$.

Nhược điểm của ma trận kè:

- ❖ Bất kể số cạnh của đồ thị là nhiều hay ít, ma trận kè luôn luôn đòi hỏi n^2 ô nhớ để lưu các phần tử ma trận, điều đó gây lãng phí bộ nhớ dẫn tới việc không thể biểu diễn được đồ thị với số đỉnh lớn.
- ❖ Với một đỉnh u bất kỳ của đồ thị, nhiều khi ta phải xét tất cả các đỉnh v khác kè với nó, hoặc xét tất cả các cạnh liên thuộc với nó. Trên ma trận kè việc đó được thực hiện bằng cách xét tất cả các đỉnh v và kiểm tra điều kiện $a[u, v] \neq 0$. Như vậy, ngay cả khi đỉnh u là **đỉnh cô lập** (không kè với đỉnh nào) hoặc **đỉnh treo** (chỉ kè với 1 đỉnh) ta cũng buộc phải xét tất cả các đỉnh và kiểm tra điều kiện trên dẫn tới lãng phí thời gian

2.2. DANH SÁCH CẠNH (EDGE LIST)

Trong trường hợp đồ thị có n đỉnh, m cạnh, ta có thể biểu diễn đồ thị dưới dạng danh sách cạnh bằng cách liệt kê tất cả các cạnh của đồ thị trong một danh sách, mỗi phần tử của danh sách là một cặp (u, v) tương ứng với một cạnh của đồ thị. (Trong trường hợp đồ thị có hướng thì mỗi cặp (u, v) tương ứng với một cung, u là đỉnh đầu và v là đỉnh cuối của cung). Danh sách được lưu trong bộ nhớ dưới dạng mảng hoặc danh sách móc nối. Ví dụ với đồ thị ở Hình 54:




Hình 54

Cài đặt trên mảng:

1	2	3	4	5	6
(1, 2)	(1, 3)	1, 5)	(2, 3)	(3, 4)	(4, 5)

Cài đặt trên danh sách móc nối:



Ưu điểm của danh sách cạnh:

- ❖ Trong trường hợp đồ thị thưa (có số cạnh tương đối nhỏ: chẳng hạn $m < 6n$), cách biểu diễn bằng danh sách cạnh sẽ tiết kiệm được không gian lưu trữ, bởi nó chỉ cần $2m$ ô nhớ để lưu danh sách cạnh.
- ❖ Trong một số trường hợp, ta phải xét tất cả các cạnh của đồ thị thì cài đặt trên danh sách cạnh làm cho việc duyệt các cạnh dễ dàng hơn. (Thuật toán Kruskal chẳng hạn)

Nhược điểm của danh sách cạnh:


- ❖ Nhược điểm cơ bản của danh sách cạnh là khi ta cần duyệt tất cả các đỉnh kè với đỉnh v nào đó của đồ thị, thì chẳng có cách nào khác là phải duyệt tất cả các cạnh, lọc ra những

cạnh có chứa đỉnh v và xét đỉnh còn lại. Điều đó khá tốn thời gian trong trường hợp đồ thị dày (nhiều cạnh).

2.3. DANH SÁCH KÈ (ADJACENCY LIST)

Để khắc phục nhược điểm của các phương pháp ma trận kè và danh sách cạnh, người ta đề xuất phương pháp biểu diễn đồ thị bằng danh sách kè. Trong cách biểu diễn này, với mỗi đỉnh v của đồ thị, ta cho tương ứng với nó một danh sách các đỉnh kè với v.

Với đồ thị $G = (V, E)$. V gồm n đỉnh và E gồm m cạnh. Có hai cách cài đặt danh sách kè phổ biến:



Hình 55

Cách 1: Dùng một mảng các đỉnh, mảng đó chia làm n đoạn, đoạn thứ i trong mảng lưu danh sách các đỉnh kè với đỉnh i: Với đồ thị ở Hình 55, danh sách kè sẽ là một mảng Adj gồm 12 phần tử:


1	2	3	4	5	6	7	8	9	10	11	12
2	3	5	1	3	1	2	4	3	5	1	4

I II III IV V

Để biết một đoạn nằm từ chỉ số nào đến chỉ số nào, ta có một mảng Head lưu vị trí riêng. $Head[i]$ sẽ bằng chỉ số đứng liền trước đoạn thứ i. Quy ước $Head[n + 1]$ bằng m. Với đồ thị bên thì mảng $Head[1..6]$ sẽ là: (0, 3, 5, 8, 10, 12)

Các phần tử $Adj[Head[i] + 1..Head[i + 1]]$ sẽ chứa các đỉnh kè với đỉnh i. Lưu ý rằng với đồ thị có hướng gồm m cung thì cấu trúc này cần phải đủ chứa m phần tử, với đồ thị vô hướng m cạnh thì cấu trúc này cần phải đủ chứa $2m$ phần tử

Cách 2: Dùng các danh sách mốc nối: Với mỗi đỉnh i của đồ thị, ta cho tương ứng với nó một danh sách mốc nối các đỉnh kè với i, có nghĩa là tương ứng với một đỉnh i, ta phải lưu lại $List[i]$ là chốt của một danh sách mốc nối. Ví dụ với đồ thị ở Hình 55, các danh sách mốc nối sẽ là:



Ưu điểm của danh sách kè:

- ❖ Đối với danh sách kè, việc duyệt tất cả các đỉnh kè với một đỉnh v cho trước là hết sức dễ dàng, cái tên “danh sách kè” đã cho thấy rõ điều này. Việc duyệt tất cả các cạnh cũng đơn giản vì một cạnh thực ra là nối một đỉnh với một đỉnh khác kè nó.

Nhược điểm của danh sách kè

- ❖ Danh sách kè yếu hơn ma trận kè ở việc kiểm tra (u, v) có phải là cạnh hay không, bởi trong cách biểu diễn này ta sẽ phải việc phải duyệt toàn bộ danh sách kè của u hay danh sách kè của v .

Đối với những thuật toán mà ta sẽ khảo sát, danh sách kè tốt hơn hẳn so với hai phương pháp biểu diễn trước. Chỉ có điều, trong trường hợp cụ thể mà ma trận kè hay danh sách cạnh **không thể hiện nhược điểm** thì ta nên dùng ma trận kè (hay danh sách cạnh) bởi cài đặt danh sách kè có phần dài dòng hơn.

2.4. NHẬN XÉT

Trên đây là nêu các cách biểu diễn đồ thị trong bộ nhớ của máy tính, còn nhập dữ liệu cho đồ thị thì có nhiều cách khác nhau, dùng cách nào thì tùy. Chẳng hạn nếu biểu diễn bằng ma trận kè mà cho nhập dữ liệu cả ma trận cấp $n \times n$ (n là số đỉnh) thì khi nhập từ bàn phím sẽ rất mất thời gian, ta cho nhập kiểu danh sách cạnh cho nhanh. Chẳng hạn mảng A ($n \times n$) là ma trận kè của một đồ thị vô hướng thì ta có thể khởi tạo ban đầu mảng A gồm toàn số 0, sau đó cho người sử dụng nhập các cạnh bằng cách nhập các cặp (i, j) ; chương trình sẽ tăng $A[i, j]$ và $A[j, i]$ lên 1. Việc nhập có thể cho kết thúc khi người sử dụng nhập giá trị $i = 0$. Ví dụ:

```
program GraphInput;
var
  A: array[1..100, 1..100] of Integer; {Ma trận kè của đồ thị}
  n, i, j: Integer;
begin
  Write('Number of vertices'); ReadLn(n);
  FillChar(A, SizeOf(A), 0);
  repeat
    Write('Enter edge (i, j) (i = 0 to exit) ');
    ReadLn(i, j); {Nhập một cặp (i, j) tương tự là nhập danh sách cạnh}
    if i <= n then
      Inc(A[i, j], 1)
    else
      Inc(A[j, i], 1)
  until i = 0;
end.
```

```
if i <> 0 then
begin {nhưng lưu trữ trong bộ nhớ lại theo kiểu ma trận kè}
  Inc(A[i, j]);
  Inc(A[j, i]);
end;
until i = 0; {Nếu người sử dụng nhập giá trị i = 0 thì dừng quá trình nhập, nếu không thì tiếp tục}
end.
```

Trong nhiều trường hợp đủ không gian lưu trữ, việc chuyển đổi từ cách biểu diễn nào đó sang cách biểu diễn khác không có gì khó khăn. Nhưng đối với thuật toán này thì làm trên ma trận kè ngắn gọn hơn, đối với thuật toán kia có thể làm trên danh sách cạnh dễ dàng hơn v.v... Do đó, với mục đích dễ hiểu, các chương trình sau này sẽ lựa chọn phương pháp biểu diễn sao cho việc cài đặt đơn giản nhất nhằm nêu bật được bản chất thuật toán. Còn trong trường hợp cụ thể bắt buộc phải dùng một cách biểu diễn nào đó khác, thì việc sửa đổi chương trình cũng không tốn quá nhiều thời gian.

§3. CÁC THUẬT TOÁN TÌM KIẾM TRÊN ĐỒ THỊ


3.1. BÀI TOÁN

Cho đồ thị $G = (V, E)$. u và v là hai đỉnh của G . Một **đường đi** (path) độ dài p từ đỉnh s đến đỉnh f là dãy $x[0..p]$ thoả mãn $x[0] = s$, $x[p] = f$ và $(x[i], x[i+1]) \in E$ với $\forall i: 0 \leq i < p$.

Đường đi nói trên còn có thể biểu diễn bởi dãy các cạnh: $(s = x[0], x[1]), (x[1], x[2]), \dots, (x[p-1], x[p] = f)$

Đỉnh u được gọi là đỉnh đầu, đỉnh v được gọi là đỉnh cuối của đường đi. Đường đi có đỉnh đầu trùng với đỉnh cuối gọi là **chu trình** (Circuit), đường đi không có cạnh nào đi qua hơn 1 lần gọi là **đường đi đơn**, tương tự ta có khái niệm **chu trình đơn**.

Ví dụ: Xét một đồ thị vô hướng và một đồ thị có hướng trong Hình 56:



Hình 56: Đồ thị và đường đi

Trên cả hai đồ thị, $(1, 2, 3, 4)$ là đường đi đơn độ dài 3 từ đỉnh 1 tới đỉnh 4. $(1, 6, 5, 4)$ không phải đường đi vì không có cạnh (cung) nối từ đỉnh 6 tới đỉnh 5.

Một bài toán quan trọng trong lý thuyết đồ thị là bài toán duyệt tất cả các đỉnh có thể đến được từ một đỉnh xuất phát nào đó. Vấn đề này đưa về một bài toán liệt kê mà yêu cầu của nó là không được bỏ sót hay lặp lại bất kỳ đỉnh nào. Chính vì vậy mà ta phải xây dựng những thuật toán cho phép duyệt một cách hệ thống các đỉnh, những thuật toán như vậy gọi là những thuật toán **tìm kiếm trên đồ thị** và ở đây ta quan tâm đến hai thuật toán cơ bản nhất: **thuật toán tìm kiếm theo chiều sâu** và **thuật toán tìm kiếm theo chiều rộng** cùng với một số ứng dụng của chúng.

Những cài đặt dưới đây là cho đơn đồ thị vô hướng, muốn làm với đồ thị có hướng hay đa đồ thị cũng không phải sửa đổi gì nhiều.


Input: file văn bản PATH.INP. Trong đó:

- ❖ Dòng 1 chứa số đỉnh n (≤ 100), số cạnh m của đồ thị, đỉnh xuất phát s , đỉnh kết thúc f cách nhau một dấu cách.
- ❖ m dòng tiếp theo, mỗi dòng có dạng hai số nguyên dương u, v cách nhau một dấu cách, thể hiện có cạnh nối đỉnh u và đỉnh v trong đồ thị.

Output: file văn bản PATH.OUT:

- ❖ Danh sách các đỉnh có thể đến được từ s

❖ Đường đi từ s tới f



PATH . INP	PATH . OUT
8 7 1 5	From 1 you can visit:
1 2	1, 2, 3, 4, 5, 6,
1 3	The path from 1 to 5:
2 3	5<-3<-2<-1
2 4	
3 5	
4 6	
7 8	

3.2. THUẬT TOÁN TÌM KIẾM THEO CHIỀU SÂU (DEPTH FIRST SEARCH)

Tư tưởng của thuật toán có thể trình bày như sau: Trước hết, mọi đỉnh x kề với S tất nhiên sẽ đến được từ S. Với mỗi đỉnh x kề với S đó thì tất nhiên những đỉnh y kề với x cũng đến được từ S... Điều đó gợi ý cho ta viết một thủ tục đệ quy DFS(u) mô tả việc duyệt từ đỉnh u bằng cách thăm đỉnh u và tiếp tục quá trình duyệt DFS(v) với v là một đỉnh chưa thăm kề với u.

Để không một đỉnh nào bị liệt kê tới hai lần, ta sử dụng kỹ thuật đánh dấu, mỗi lần thăm một đỉnh, ta đánh dấu đỉnh đó lại để các bước duyệt đệ quy kế tiếp không duyệt lại đỉnh đó nữa.

Để lưu lại đường đi từ đỉnh xuất phát s, trong thủ tục DFS(u), trước khi gọi đệ quy DFS(v) với v là một đỉnh kề với u mà chưa đánh dấu, ta lưu lại vết đường đi từ u tới v bằng cách đặt Trace[v] := u, tức là Trace[v] lưu lại đỉnh liền trước v trong đường đi từ s tới v. Khi quá trình tìm kiếm theo chiều sâu kết thúc, đường đi từ s tới f sẽ là:

$$f \leftarrow p[1] = \text{Trace}[f] \leftarrow p[2] = \text{Trace}[p[1]] \leftarrow \dots \leftarrow s.$$

```

procedure DFS(u∈V);
begin
  Free[u] := False; {Free[u] = False ⇔ u đã thăm}
  for (forall v ∈ V: Free[v] and ((u, v) ∈ E)) do {Duyệt mọi đỉnh v chưa thăm kề với u}
    begin
      Trace[v] := u; {Lưu vết đường đi, đỉnh liền trước v trên đường đi từ s tới v là u}
      DFS(v); {Gọi đệ quy duyệt tương tự đối với v}
    end;
  end;

  begin {Chương trình chính}
    {Nhập dữ liệu: đồ thị, đỉnh xuất phát S, đỉnh đích F};
    for (forall v ∈ V) do Free[v] := True; {Đánh dấu mọi đỉnh đều chưa thăm}
    DFS(S);
    {Thông báo từ s có thể thăm được những đỉnh v mà Free[v] = False};
    if Free[f] then {s đi tới được f}
      {Truy theo vết từ f để tìm đường đi từ s tới f};
  end.

```

Trong cài đặt cụ thể, ta không cần mảng đánh dấu Free[1..n] mà dùng luôn mảng Trace[1..n] để đánh dấu, khởi tạo các phần tử Trace[s] := -1 và Trace[v] := 0 với $\forall v \neq s$. Điều kiện để một đỉnh v chưa thăm là Trace[v] = 0, mỗi khi từ đỉnh u thăm đỉnh v, phép gán Trace[v] := u sẽ kiêm luôn công việc đánh dấu v đã thăm.

P_4_03_1.PAS * Thuật toán tìm kiếm theo chiều sâu

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-2^31..2^31 - 1]*)
program Depth_First_Search;

```

```

const
  InputFile = 'PATH.INP';
  OutputFile = 'PATH.OUT';
  max = 100;
var
  a: array[1..max, 1..max] of Boolean;
  Trace: array[1..max] of Integer;
  n, s, f: Integer;

procedure Enter; {Nhập dữ liệu}
var
  i, u, v, m: Integer;
  fi: Text;
begin
  Assign(fi, InputFile); Reset(fi);
  FillChar(a, SizeOf(a), False);
  ReadLn(fi, n, m, s, f);
  for i := 1 to m do
    begin
      ReadLn(fi, u, v);
      a[u, v] := True;
      a[v, u] := True; {Đó thị vô hướng nên cạnh (u, v) cũng là cạnh (v, u)}
    end;
  Close(fi);
end;

procedure DFS(u: Integer); {Thuật toán tìm kiếm theo chiều sâu bắt đầu từ u}
var
  v: Integer;
begin
  for v := 1 to n do
    if a[u, v] and (Trace[v] = 0) then {Duyệt ∀v chưa thăm kè với u}
      begin
        Trace[v] := u; {Lưu vết đường đi cũng là đánh dấu v đã thăm}
        DFS(v); {Tìm kiếm theo chiều sâu bắt đầu từ v}
      end;
end;

procedure Result; {In kết quả}
var
  fo: Text;
  v: Integer;
begin
  Assign(fo, OutputFile); Rewrite(fo);
  Writeln(fo, 'From ', s, ' you can visit: ');
  for v := 1 to n do
    if Trace[v] <> 0 then Write(fo, v, ', '); {In ra những đỉnh đến được từ s}
  Writeln(fo);
  Writeln(fo, 'The path from ', s, ' to ', f, ': ');
  if Trace[f] = 0 then {Nếu Trace[f] = 0 thì s không thể tới được f}
    Writeln(fo, 'not found')
  else {s tới được f}
    begin
      while f <> s do {Truy vết đường đi}
        begin
          Write(fo, f, '<-');
          f := Trace[f];
        end;
      Writeln(fo, s);
    end;
  Close(fo);
end;

```


```

begin
  Enter;
  FillChar(Trace, SizeOf(Trace), 0); {Mọi đỉnh đều chưa thăm}
  Trace[s] := -1; {Ngoại trừ s đã thăm}
  DFS(s);
  Result;
end.

```

Nhận xét:


- ❖ Vì có kỹ thuật đánh dấu, nên thủ tục DFS sẽ được gọi $\leq n$ lần (n là số đỉnh)
- ❖ Có thể có nhiều đường đi từ s tới f , nhưng thuật toán DFS luôn trả về một đường đi có thứ tự từ điển nhỏ nhất.
- ❖ Quá trình tìm kiếm theo chiều sâu cho ta một cây DFS gốc s . Quan hệ cha-con trên cây được định nghĩa là: nếu từ đỉnh u tới thăm đỉnh v ($DFS(u)$ gọi $DFS(v)$) thì u là nút cha của nút v . Hình 57 là đồ thị và cây DFS tương ứng với đỉnh xuất phát $s = 1$.



Hình 57: Cây DFS

3.3. THUẬT TOÁN TÌM KIẾM THEO CHIỀU RỘNG (BREADTH FIRST SEARCH)

Cơ sở của phương pháp cài đặt này là “lập lịch” duyệt các đỉnh. Việc thăm một đỉnh sẽ lên lịch duyệt các đỉnh kề nó sao cho thứ tự duyệt là ưu tiên chiều rộng (đỉnh nào gần S hơn sẽ được duyệt trước). Ví dụ: Bắt đầu ta thăm đỉnh S . Việc thăm đỉnh S sẽ phát sinh thứ tự duyệt những đỉnh ($x[1], x[2], \dots, x[p]$) kề với S (những đỉnh gần S nhất). Khi thăm đỉnh $x[1]$ sẽ lại phát sinh yêu cầu duyệt những đỉnh ($u[1], u[2], \dots, u[q]$) kề với $x[1]$. Nhưng rõ ràng các đỉnh u này “xa” S hơn những đỉnh x nên chúng chỉ được duyệt khi tất cả những đỉnh x đã duyệt xong. Tức là thứ tự duyệt đỉnh sau khi đã thăm $x[1]$ sẽ là: ($x[2], x[3], \dots, x[p], u[1], u[2], \dots, u[q]$).



Giả sử ta có một danh sách chứa những đỉnh đang “chờ” thăm. Tại mỗi bước, ta thăm một đỉnh đầu danh sách và cho những đỉnh chưa “xếp hàng” kè với nó xếp hàng thêm vào cuối danh sách. Chính vì nguyên tắc đó nên danh sách chứa những đỉnh đang chờ sẽ được tổ chức dưới dạng hàng đợi (Queue).

Nếu ta có Queue là một hàng đợi với thủ tục Push(v) để đẩy một đỉnh v vào hàng đợi và hàm Pop trả về một đỉnh lấy ra từ hàng đợi thì mô hình của giải thuật có thể viết như sau:

```

for ( $\forall v \in V$ ) do Free[v] := True;
Free[s] := False; {Khởi tạo ban đầu chỉ có đỉnh S bị đánh dấu}
Queue := Ø; Push(s); {Khởi tạo hàng đợi ban đầu chỉ gồm một đỉnh s}
repeat {Lặp tới khi hàng đợi rỗng}
    u := Pop; {Lấy từ hàng đợi ra một đỉnh u}
    for ( $\forall v \in V: Free[v]$  and  $((u, v) \in E)$ ) do {Xét những đỉnh v kề u chưa bị đưa vào hàng đợi}
        begin
            Trace[v] := u; {Lưu vết đường đi}
            Free[v] := False; {Đánh dấu v}
            Push(v); {Đẩy v vào hàng đợi}
        end;
    until Queue = Ø;
{Thông báo từ s có thể thăm được những đỉnh v mà Free[v] = False};
if Free[f] then {s đã tới được f}
    {Truy theo vết từ f để tìm đường đi từ s tới f};

```

Tương tự như thuật toán tìm kiếm theo chiều sâu, ta có thể dùng mảng Trace[1..n] kiêm luôn chức năng đánh dấu.

P_4_03_2.PAS * Thuật toán tìm kiếm theo chiều rộng

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Breadth_First_Search;
const
    InputFile = 'PATH.INP';
    OutputFile = 'PATH.OUT';
    max = 100;
var
    a: array[1..max, 1..max] of Boolean;
    Trace: array[1..max] of Integer;
    n, s, f: Integer;

procedure Enter; {Nhập dữ liệu}
var
    i, u, v, m: Integer;
    fi: Text;
begin
    Assign(fi, InputFile); Reset(fi);
    FillChar(a, SizeOf(a), False);
    ReadLn(fi, n, m, s, f);
    for i := 1 to m do
        begin
            ReadLn(fi, u, v);
            a[u, v] := True;
            a[v, u] := True;
        end;
    Close(fi);
end;

procedure BFS; {Thuật toán tìm kiếm theo chiều rộng}
var
    Queue: array[1..max] of Integer;
    Front, Rear, u, v: Integer;

```

```

begin
  Front := 1; Rear := 1; {front: chỉ số đầu hàng đợi, rear: chỉ số cuối hàng đợi}
  Queue[1] := s; {Khởi tạo hàng đợi ban đầu chỉ gồm một đỉnh s}
  FillChar(Trace, SizeOf(Trace), 0); {Các đỉnh đều chưa đánh dấu}
  Trace[s] := -1; {Ngoại trừ đỉnh s}
  repeat {Lặp tới khi hàng đợi rỗng}
    u := Queue[Front]; Inc(Front); {Lấy từ hàng đợi ra một đỉnh u}
    for v := 1 to n do
      if a[u, v] and (Trace[v] = 0) then {Xét những đỉnh v chưa thăm kề với u}
        begin
          Inc(Rear); Queue[Rear] := v; {Đẩy v vào hàng đợi}
          Trace[v] := u; {Lưu vết đường đi, cũng là đánh dấu luôn}
        end;
    until Front > Rear;
  end;


procedure Result; {In kết quả}
var
  fo: Text;
  v: Integer;
begin
  Assign(fo, OutputFile); Rewrite(fo);
  Writeln(fo, 'From ', s, ' you can visit: ');
  for v := 1 to n do
    if Trace[v] <> 0 then Write(fo, v, ', ');
  Writeln(fo);
  Writeln(fo, 'The path from ', s, ' to ', f, ': ');
  if Trace[f] = 0 then
    Writeln(fo, 'not found')
  else
    begin
      while f <> s do
        begin
          Write(fo, f, '<-');
          f := Trace[f];
        end;
      Writeln(fo, s);
    end;
  Close(fo);
end;

begin
  Enter;
  BFS;
  Result;
end.

```

Nhận xét:

- ❖ Có thể có nhiều đường đi từ s tới f nhưng thuật toán BFS luôn trả về một đường đi ngắn nhất (theo nghĩa đi qua ít cạnh nhất).
- ❖ Quá trình tìm kiếm theo chiều rộng cho ta một cây BFS gốc s. Quan hệ cha - con trên cây được định nghĩa là: nếu từ đỉnh u tới thăm đỉnh v thì u là nút cha của nút v. Hình 58 là ví dụ về cây BFS.

**Hình 58: Cây BFS**

3.4. ĐỘ PHÚC TẠP TÍNH TOÁN CỦA BFS VÀ DFS

Quá trình tìm kiếm trên đồ thị bắt đầu từ một đỉnh có thể thăm tất cả các đỉnh còn lại, khi đó cách biểu diễn đồ thị có ảnh hưởng lớn tới chi phí về thời gian thực hiện giải thuật:

- ❖ Trong trường hợp ta biểu diễn đồ thị bằng danh sách kè, cả hai thuật toán BFS và DFS đều có độ phức tạp tính toán là $O(n + m) = O(\max(n, m))$. Đây là cách cài đặt tốt nhất.
- ❖ Nếu ta biểu diễn đồ thị bằng ma trận kè như ở trên thì độ phức tạp tính toán trong trường hợp này là $O(n + n^2) = O(n^2)$.
- ❖ Nếu ta biểu diễn đồ thị bằng danh sách cạnh, thao tác duyệt những đỉnh kè với đỉnh u sẽ dẫn tới việc phải duyệt qua toàn bộ danh sách cạnh, đây là cài đặt tồi nhất, nó có độ phức tạp tính toán là $O(n.m)$.

Bài tập

Bài 1

Mê cung hình chữ nhật kích thước $m \times n$ gồm các ô vuông đơn vị. Trên mỗi ô ghi một trong ba ký tự:

- ❖ O: Nếu ô đó an toàn
- ❖ X: Nếu ô đó có cạm bẫy
- ❖ E: Nếu là ô có một nhà thám hiểm đang đứng.

Duy nhất chỉ có 1 ô ghi chữ E. Nhà thám hiểm có thể từ một ô đi sang một trong số các ô chung cạnh với ô đang đứng. Một cách đi thoát khỏi mê cung là một hành trình đi qua các ô an toàn ra một ô biên. Hãy chỉ giúp cho nhà thám hiểm một hành trình thoát ra khỏi mê cung.

Bài 2


Có nhiều cách dựng cây gốc s chứa tất cả các đỉnh đến được từ s thoả mãn: nếu u là nút cha của nút v trên cây thì (u, v) phải là cạnh của G. Cây BFS và cây DFS chỉ là hai trường hợp đặc biệt. Ta đã biết cây BFS là cây thấp nhất, vậy phải chăng cây DFS là cây cao nhất trong số các cây này?

§4. TÍNH LIÊN THÔNG CỦA ĐỒ THỊ

4.1. ĐỊNH NGHĨA


4.1.1. Đôi với đồ thị vô hướng $G = (V, E)$

G gọi là **liên thông** (connected) nếu luôn tồn tại đường đi giữa mọi cặp đỉnh phân biệt của đồ thị. Nếu G không liên thông thì chắc chắn nó sẽ là hợp của hai hay nhiều đồ thị con* liên thông, các đồ thị con này đôi một không có đỉnh chung. Các đồ thị con liên thông rời nhau như vậy được gọi là các thành phần liên thông của đồ thị đang xét (Xem ví dụ).



Hình 59: Đồ thị G và các thành phần liên thông G_1, G_2, G_3 của nó

Đôi khi, việc xoá đi một đỉnh và tất cả các cạnh liên thuộc với nó sẽ tạo ra một đồ thị con mới có nhiều thành phần liên thông hơn đồ thị ban đầu, các đỉnh như thế gọi là **đỉnh cắt** hay **điểm khớp**. Hoàn toàn tương tự, những cạnh mà khi ta bỏ nó đi sẽ tạo ra một đồ thị có nhiều thành phần liên thông hơn so với đồ thị ban đầu được gọi là **cạnh cắt** hay **cầu**.




Hình 60: Khớp và cầu

4.1.2. Đôi với đồ thị có hướng $G = (V, E)$

Có hai khái niệm về tính liên thông của đồ thị có hướng tùy theo chúng ta có quan tâm tới hướng của các cung không.

G gọi là **liên thông mạnh** (Strongly connected) nếu luôn tồn tại đường đi (theo các cung định hướng) giữa hai đỉnh bất kỳ của đồ thị, G gọi là **liên thông yếu** (weakly connected) nếu phiên bản vô hướng của nó là đồ thị liên thông.

* Đồ thị $G = (V, E)$ là con của đồ thị $G' = (V', E')$ nếu G là đồ thị có $V \subseteq V'$ và $E \subseteq E'$



Hình 61: Liên thông mạnh và liên thông yếu

4.2. TÍNH LIÊN THÔNG TRONG ĐỒ THỊ VÔ HƯỚNG

Một bài toán quan trọng trong lý thuyết đồ thị là bài toán kiểm tra tính liên thông của đồ thị vô hướng hay tổng quát hơn: Bài toán liệt kê các thành phần liên thông của đồ thị vô hướng.

Giả sử đồ thị vô hướng $G = (V, E)$ có n đỉnh đánh số $1, 2, \dots, n$.

Để liệt kê các thành phần liên thông của G phương pháp cơ bản nhất là:

Đánh dấu đỉnh 1 và những đỉnh có thể đến từ 1 , thông báo những đỉnh đó thuộc thành phần liên thông thứ nhất.

Nếu tất cả các đỉnh đều đã bị đánh dấu thì G là đồ thị liên thông, nếu không thì sẽ tồn tại một đỉnh v nào đó chưa bị đánh dấu, ta sẽ đánh dấu v và các đỉnh có thể đến được từ v , thông báo những đỉnh đó thuộc thành phần liên thông thứ hai.

Và cứ tiếp tục như vậy cho tới khi tất cả các đỉnh đều đã bị đánh dấu

```
procedure Scan(u)
begin
  (Dùng BFS hoặc DFS liệt kê và đánh dấu những đỉnh có thể đến được từ u);
end;
```

```
begin
  for  $\forall v \in V$  do (khởi tạo  $v$  chưa đánh dấu);
  Count := 0;
  for  $u := 1$  to  $n$  do
    if  $\langle u$  chưa đánh dấu  $\rangle$  then
      begin
        Count := Count + 1;
        WriteLn('Thành phần liên thông thứ ', Count, ' gồm các đỉnh : ');
        Scan(u);
      end;
  end.
```


Với thuật toán liệt kê các thành phần liên thông như thế này, thì độ phức tạp tính toán của nó đúng bằng độ phức tạp tính toán của thuật toán tìm kiếm trên đồ thị trong thủ tục Scan.

4.3. ĐỒ THỊ ĐẦY ĐỦ VÀ THUẬT TOÁN WARSHALL

4.3.1. Định nghĩa:

Đồ thị đầy đủ với n đỉnh, ký hiệu K_n , là một đơn đồ thị vô hướng mà giữa hai đỉnh bất kỳ của nó đều có cạnh nối.

Đồ thị đầy đủ K_n có đúng: $\binom{n}{2} = \frac{n(n-1)}{2}$ cạnh và bậc của mọi đỉnh đều bằng $n - 1$.

**Hình 62: Đồ thị đầy đủ****4.3.2. Bao đóng đồ thị:**


Với đồ thị $G = (V, E)$, người ta xây dựng đồ thị $G' = (V, E')$ cũng gồm những đỉnh của G còn các cạnh xây dựng như sau: (ở đây quy ước giữa u và v luôn có đường đi)

Giữa đỉnh u và v của G' có cạnh nối \Leftrightarrow Giữa đỉnh u và v của G có đường đi

Đồ thị G' xây dựng như vậy được gọi là bao đóng của đồ thị G .

Từ định nghĩa của đồ thị đầy đủ, ta dễ dàng suy ra một đồ thị đầy đủ bao giờ cũng liên thông và từ định nghĩa đồ thị liên thông, ta cũng dễ dàng suy ra được:

- ❖ Một đơn đồ thị vô hướng là liên thông nếu và chỉ nếu bao đóng của nó là đồ thị đầy đủ
- ❖ Một đơn đồ thị vô hướng có k thành phần liên thông nếu và chỉ nếu bao đóng của nó có k thành phần liên thông đầy đủ.

**Hình 63: Đơn đồ thị vô hướng và bao đóng của nó**

Bởi việc kiểm tra một đồ thị có phải đồ thị đầy đủ hay không có thể thực hiện khá dễ dàng (đếm số cạnh chẵng hạn) nên người ta nảy ra ý tưởng có thể kiểm tra tính liên thông của đồ thị thông qua việc kiểm tra tính đầy đủ của bao đóng. Vấn đề đặt ra là phải có thuật toán xây dựng bao đóng của một đồ thị cho trước và một trong những thuật toán đó là:

4.3.3. Thuật toán Warshall

Thuật toán Warshall - gọi theo tên của Stephen Warshall, người đã mô tả thuật toán này vào năm 1960, đôi khi còn được gọi là thuật toán Roy-Warshall vì Roy cũng đã mô tả thuật toán này vào năm 1959. Thuật toán đó có thể mô tả rất gọn:

Với đơn đồ thị vô hướng G , với mọi đỉnh k xét theo thứ tự từ 1 tới n , ta xét tất cả các cặp đỉnh (u, v) : nếu có cạnh nối (u, k) và cạnh nối (k, v) thì ta tự nối thêm cạnh (u, v) nếu nó chưa có.

Tư tưởng này dựa trên một quan sát đơn giản như sau: Nếu từ u có đường đi tới k và từ k lại có đường đi tới v thì tất nhiên từ u sẽ có đường đi tới v .

Với n là số đỉnh của đồ thị và $A = \{a[i, j]\}$ là ma trận kề biểu diễn đồ thị, ta có thể viết thuật toán Warshall như sau:


```
for k := 1 to n do
    for u := 1 to n do
        for v := 1 to n do
            a[u, v] := a[u, v] or a[u, k] and a[k, v];
```

Việc chứng minh tính đúng đắn của thuật toán đòi hỏi phải lật lại các lý thuyết về bao đóng bắc cầu và quan hệ liên thông, ta sẽ không trình bày ở đây. Có nhận xét rằng tuy thuật toán Warshall rất dễ cài đặt nhưng độ phức tạp tính toán của thuật toán này khá lớn ($O(n^3)$).

Dưới đây, ta sẽ thử cài đặt thuật toán Warshall tìm bao đóng của đơn đồ thị vô hướng sau đó đếm số thành phần liên thông của đồ thị:

Việc cài đặt thuật toán sẽ qua những bước sau:


- ❖ Nhập ma trận kề A của đồ thị (Lưu ý ở đây $A[v, v]$ luôn được coi là True với $\forall v$)
- ❖ Dùng thuật toán Warshall tìm bao đóng, khi đó A là ma trận kề của bao đóng đồ thị
- ❖ Dựa vào ma trận kề A , đỉnh 1 và những đỉnh kề với nó sẽ thuộc thành phần liên thông thứ nhất; với đỉnh u nào đó không kề với đỉnh 1, thì u cùng với những đỉnh kề nó sẽ thuộc thành phần liên thông thứ hai; với đỉnh v nào đó không kề với cả đỉnh 1 và đỉnh u , thì v cùng với những đỉnh kề nó sẽ thuộc thành phần liên thông thứ ba v.v...



Input: file văn bản CONNECT.INP

- ❖ Dòng 1: Chứa số đỉnh n (≤ 100) và số cạnh m của đồ thị cách nhau ít nhất một dấu cách
- ❖ m dòng tiếp theo, mỗi dòng chứa một cặp số u và v cách nhau ít nhất một dấu cách tượng trưng cho một cạnh (u, v)

Output: file văn bản CONNECT.OUT, liệt kê các thành phần liên thông



CONNECT.INP	CONNECT.OUT
12 9	Connected Component 1:
1 3	1, 2, 3, 4, 5,
1 4	Connected Component 2:
1 5	6, 7, 8,
2 4	Connected Component 3:
6 7	9, 10, 11, 12,
6 8	
9 10	
9 11	
11 12	

```

P_4_04_1.PAS * Thuật toán Warshall liệt kê các thành phần liên thông
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Connectivity;
const
  InputFile = 'CONNECT.INP';
  OutputFile = 'CONNECT.OUT';
  max = 100;
var
  a: array[1..max, 1..max] of Boolean; {Ma trận kề của đồ thị}
  Free: array[1..max] of Boolean; {Free[v] = True ⇔ v chưa được liệt kê vào thành phần liên thông nào}
  k, u, v, n: Integer;
  Count: Integer;
  fo: Text;
begin
  procedure Enter; {Nhập đồ thị}
  var
    i, u, v, m: Integer;
    fi: Text;
  begin
    FillChar(a, SizeOf(a), False);
    Assign(fi, InputFile); Reset(fi);
    ReadLn(fi, n, m);
    for v := 1 to n do a[v, v] := True; {Đi nhiên từ v có đường đi đến chính v}
    for i := 1 to m do
      begin
        ReadLn(fi, u, v);
        a[u, v] := True;
        a[v, u] := True;
      end;
    Close(fi);
  end;

  begin
    Enter;
    {Thuật toán Warshall}
    for k := 1 to n do
      for u := 1 to n do
        for v := 1 to n do
          a[u, v] := a[u, v] or a[u, k] and a[k, v];
    Assign(fo, OutputFile); Rewrite(fo);
    Count := 0;
    FillChar(Free, n, True); {Mọi đỉnh đều chưa được liệt kê vào thành phần liên thông nào}
    for u := 1 to n do
      if Free[u] then {Với một đỉnh u chưa được liệt kê vào thành phần liên thông nào}
        begin
          Inc(Count);
          WriteLn(fo, 'Connected Component ', Count, ': ');
          for v := 1 to n do
            if a[u, v] then {Xét những đỉnh kề u (trên bao đóng)}
              begin
                Write(fo, v, ', ');
                Free[v] := False; {Liệt kê đỉnh nào đánh dấu đỉnh đó}
              end;
          WriteLn(fo);
        end;
    Close(fo);
  end.
end.

```

4.4. CÁC THÀNH PHẦN LIÊN THÔNG MẠNH

Đối với đồ thị có hướng, người ta quan tâm đến bài toán kiểm tra tính liên thông mạnh, hay tổng quát hơn: Bài toán liệt kê các thành phần liên thông mạnh của đồ thị có hướng. Đối với

bài toán đó ta có một phương pháp khá hữu hiệu dựa trên thuật toán tìm kiếm theo chiều sâu Depth First Search.

4.4.1. Phân tích

Thêm vào đồ thị một đỉnh x và nối x với tất cả các đỉnh còn lại của đồ thị bằng các cung định hướng. Khi đó quá trình tìm kiếm theo chiều sâu bắt đầu từ x có thể coi như một quá trình xây dựng cây tìm kiếm theo chiều sâu (cây DFS) gốc x.


```
procedure Visit(uεV);
begin
    {Thêm u vào cây tìm kiếm DFS};
    for (Vv: (v ∈ cây DFS) and ((u, v) ∈ E)) do Visit(v);
end;

begin
    {Thêm vào đồ thị đỉnh x và các cung định hướng (x, v) với mọi v};
    {Khởi tạo cây tìm kiếm DFS := Ø};
    Visit(x);
end.
```

Để ý thủ tục thăm đỉnh đệ quy Visit(u). Thủ tục này xét tất cả những đỉnh v nối từ u, nếu v chưa được thăm thì đi theo cung đó thăm v, tức là bổ sung cung (u, v) vào cây tìm kiếm DFS. Nếu v **đã thăm** thì có ba khả năng xảy ra đối với vị trí của u và v trong cây tìm kiếm DFS:

- ❖ v là tiền bối (ancestor - tổ tiên) của u, tức là v được thăm trước u và thủ tục Visit(u) do dây chuyền đệ quy từ thủ tục Visit(v) gọi tới. Cung (u, v) khi đó được gọi là **cung ngược** (Back edge)
- ❖ v là hậu duệ (descendant - con cháu) của u, tức là u được thăm trước v, nhưng thủ tục Visit(u) sau khi tiền đệ quy theo một hướng khác đã gọi Visit(v) rồi. Nên khi dây chuyền đệ quy lùi lại về thủ tục Visit(u) sẽ thấy v là đã thăm nên không thăm lại nữa. Cung (u, v) khi đó gọi là **cung xuôi** (Forward edge).
- ❖ v thuộc một nhánh của cây DFS đã duyệt trước đó, tức là sẽ có một đỉnh w được thăm trước cả u và v. Thủ tục Visit(w) gọi trước sẽ rẽ theo một nhánh nào đó thăm v trước, rồi khi lùi lại, rẽ sang một nhánh khác thăm u. Cung (u, v) khi đó gọi là **cung chéo** (Cross edge)

(Rất tiếc là từ điển thuật ngữ tin học Anh-Việt quá nghèo nàn nên không thể tìm ra những từ tương đương với các thuật ngữ ở trên. Ta có thể hiểu qua các ví dụ).



Hình 64: Ba dạng cung ngoài cây DFS

Ta nhận thấy một đặc điểm của thuật toán tìm kiếm theo chiều sâu, thuật toán không chỉ duyệt qua các đỉnh, nó còn duyệt qua tất cả những cung nữa. Ngoài những cung nằm trên cây tìm kiếm, những cung còn lại có thể chia làm ba loại: cung ngược, cung xuôi, cung chéo.

4.4.2. Cây tìm kiếm DFS và các thành phần liên thông mạnh

Định lý 1: Nếu a, b là hai đỉnh thuộc thành phần liên thông mạnh C thì với mọi đường đi từ a tới b cũng như từ b tới a . Tất cả đỉnh trung gian trên đường đi đó đều phải thuộc C .

Chứng minh

Nếu a và b là hai đỉnh thuộc C thì tức là có một đường đi từ a tới b và một đường đi khác từ b tới a . Suy ra với một đỉnh v nằm trên đường đi từ a tới b thì a tới v được v , v tới b mà b có đường tới a nên v cũng tới a . Vậy v nằm trong thành phần liên thông chứa a tức là $v \in C$. Tương tự với một đỉnh nằm trên đường đi từ b tới a .

Định lý 2: Với một thành phần liên thông mạnh C bất kỳ, sẽ tồn tại một đỉnh $r \in C$ sao cho mọi đỉnh của C đều thuộc nhánh DFS gốc r .

Chứng minh: Trước hết, nhắc lại một thành phần liên thông mạnh là một đồ thị con liên thông mạnh của đồ thị ban đầu thỏa mãn tính chất tối đại tức là việc thêm vào thành phần đó một tập hợp đỉnh khác sẽ làm mất đi tính liên thông mạnh.

Trong số các đỉnh của C , chọn r là đỉnh được thăm đầu tiên theo thuật toán tìm kiếm theo chiều sâu. Ta sẽ chứng minh C nằm trong nhánh DFS gốc r . Thật vậy: với một đỉnh v bất kỳ của C , do C liên thông mạnh nên phải tồn tại một đường đi từ r tới v :

$$(r = x[0], x[1], \dots, x[k] = v)$$

Từ định lý 1, tất cả các đỉnh $x[1], x[2], \dots, x[k]$ đều thuộc C nên chúng sẽ phải thăm sau đỉnh r . Khi thủ tục $\text{Visit}(r)$ được gọi thì tất cả các đỉnh $x[1], x[2], \dots, x[k] = v$ đều chưa thăm; vì thủ tục $\text{Visit}(r)$ sẽ liệt kê tất cả những đỉnh chưa thăm đến được từ r bằng cách xây dựng nhánh gốc r của cây DFS, nên các đỉnh $x[1], x[2], \dots, x[k] = v$ sẽ thuộc nhánh gốc r của cây DFS. Bởi chọn v là đỉnh bất kỳ trong C nên ta có điều phải chứng minh.

Đỉnh r trong chứng minh định lý - đỉnh thăm trước tất cả các đỉnh khác trong C - gọi là **chốt** của thành phần C . Mỗi thành phần liên thông mạnh có duy nhất một chốt. Xét về vị trí trong cây tìm kiếm DFS, chốt của một thành phần liên thông là đỉnh nằm cao nhất so với các đỉnh

khác thuộc thành phần đó, hay nói cách khác: là tiền bối của tất cả các đỉnh thuộc thành phần đó.

Định lý 3: Luôn tìm được đỉnh chót a thoả mãn: Quá trình tìm kiếm theo chiều sâu bắt đầu từ a không thăm được bất kỳ một chót nào khác. (Tức là nhánh DFS gốc a không chứa một chót nào ngoài a) chẳng hạn ta chọn a là chót được thăm sau cùng trong một dây chuyền đệ quy hoặc chọn a là chốt thăm sau tất cả các chốt khác. Với chốt a như vậy thì các đỉnh thuộc nhánh DFS gốc a chính là thành phần liên thông mạnh chứa a.


Chứng minh: Với mọi đỉnh v nằm trong nhánh DFS gốc a, xét b là chót của thành phần liên thông mạnh chứa v. Ta sẽ chứng minh $a \equiv b$. Thật vậy, theo định lý 2, v phải nằm trong nhánh DFS gốc b. Vậy v nằm trong cả nhánh DFS gốc a và nhánh DFS gốc b. Giả sử phản chứng rằng $a \neq b$ thì sẽ có hai khả năng xảy ra:

- ❖ **Khả năng 1:** Nhánh DFS gốc a chứa nhánh DFS gốc b, có nghĩa là thủ tục Visit(b) sẽ do thủ tục Visit(a) gọi tới, điều này mâu thuẫn với giả thiết rằng a là chót mà quá trình tìm kiếm theo chiều sâu bắt đầu từ a không thăm một chót nào khác.
- ❖ **Khả năng 2:** Nhánh DFS gốc a nằm trong nhánh DFS gốc b, có nghĩa là a nằm trên một đường đi từ b tới v. Do b và v thuộc cùng một thành phần liên thông mạnh nên theo định lý 1, a cũng phải thuộc thành phần liên thông mạnh đó. Vậy thì thành phần liên thông mạnh này có hai chốt a và b. Điều này vô lý.

Theo định lý 2, ta đã có thành phần liên thông mạnh chứa a nằm trong nhánh DFS gốc a, theo chứng minh trên ta lại có: Mọi đỉnh trong nhánh DFS gốc a nằm trong thành phần liên thông mạnh chứa a. Kết hợp lại được: Nhánh DFS gốc a chính là thành phần liên thông mạnh chứa a.

4.4.3. Thuật toán Tarjan (R.E.Tarjan - 1972)

Chọn u là chốt mà từ đó quá trình tìm kiếm theo chiều sâu không thăm thêm bất kỳ một chốt nào khác, chọn lấy thành phần liên thông mạnh thứ nhất là nhánh DFS gốc u. Sau đó loại bỏ nhánh DFS gốc u ra khỏi cây DFS, lại tìm thấy một đỉnh chốt v khác mà nhánh DFS gốc v không chứa chốt nào khác, lại chọn lấy thành phần liên thông mạnh thứ hai là nhánh DFS gốc v. Tương tự như vậy cho thành phần liên thông mạnh thứ ba, thứ tư, v.v... Có thể hình dung thuật toán Tarjan “bẻ” cây DFS tại vị trí các chốt để được các nhánh rời rạc, mỗi nhánh là một thành phần liên thông mạnh.



Hình 65: Thuật toán Tarjan “bé” cây DFS

Trình bày dài dòng như vậy, nhưng điều quan trọng nhất bây giờ mới nói tới: **Làm thế nào kiểm tra một đỉnh v nào đó có phải là chốt hay không ?**

Hãy để ý nhánh DFS gốc ở đỉnh r nào đó.

Nhận xét 1: Nếu như từ các đỉnh thuộc nhánh gốc r này không có cung ngược hay cung chéo nào đi ra khỏi nhánh đó thì r là chốt. Điều này dễ hiểu bởi như vậy có nghĩa là từ r, đi theo các cung của đồ thị thì chỉ đến được những đỉnh thuộc nhánh đó mà thôi. Vậy:

Thành phần liên thông mạnh chứa r \subseteq Tập các đỉnh có thể đến từ r = Nhánh DFS gốc r nên r là chốt.

Nhận xét 2: Nếu từ một đỉnh v nào đó của nhánh DFS gốc r có một cung ngược tới một đỉnh w là tiền bối của r, thì r không là chốt. Thật vậy: do có chu trình ($w \rightarrow r \rightarrow v \rightarrow w$) nên w, r, v thuộc cùng một thành phần liên thông mạnh. Mà w được thăm trước r, điều này mâu thuẫn với cách xác định chốt (Xem lại định lý 2)

Nhận xét 3: Vẫn đề phức tạp gấp phải ở đây là nếu từ một đỉnh v của nhánh DFS gốc r, có một cung chéo đi tới một nhánh khác. Ta sẽ thiết lập giải thuật liệt kê thành phần liên thông mạnh ngay trong thủ tục Visit(u), khi mà đỉnh u đã **đã duyệt xong**, tức là khi các đỉnh khác của nhánh DFS gốc u đều đã thăm và quá trình thăm đệ quy lùi lại về Visit(u). Nếu như u là chốt, ta thông báo nhánh DFS gốc u là thành phần liên thông mạnh chứa u và loại ngay các đỉnh thuộc thành phần đó khỏi đồ thị cũng như khỏi cây DFS. Có thể chứng minh được tính đúng đắn của phương pháp này, bởi nếu nhánh DFS gốc u chứa một chốt u' khác thì u' phải duyệt xong trước u và cả nhánh DFS gốc u' đã bị loại bỏ rồi. Hơn nữa còn có thể chứng minh được rằng, khi thuật toán tiên hành như trên thì nếu như **từ một đỉnh v của một nhánh DFS gốc r có một cung chéo đi tới một nhánh khác thì r không là chốt**.

Để chứng tỏ điều này, ta dựa vào tính chất của cây DFS: cung chéo sẽ nối từ một nhánh tới nhánh thăm trước đó, chứ không bao giờ có cung chéo đi tới nhánh thăm sau. Giả sử có cung chéo (v, v') đi từ $v \in$ nhánh DFS gốc r tới $v' \notin$ nhánh DFS gốc r, gọi r' là chốt của thành phần

liên thông chứa v' . Theo tính chất trên, v' phải thăm trước r , suy ra **r' cũng phải thăm trước r** . Có hai khả năng xảy ra:

- ❖ Nếu r' thuộc nhánh DFS đã duyệt trước r thì r' sẽ được duyệt xong trước khi thăm r , tức là khi thăm r và cả sau này khi thăm v thì nhánh DFS gốc r' đã bị huỷ, cung chéo (v, v') sẽ không được tính đến nữa.
- ❖ Nếu r' là tiền bối của r thì ta có r' đến được r , v nằm trong nhánh DFS gốc r nên r đến được v , v đến được v' vì (v, v') là cung, v' lại đến được r' bởi r' là chốt của thành phần liên thông mạnh chứa v' . Ta thiết lập được chu trình $(r' \rightarrow r \rightarrow v \rightarrow v' \rightarrow r')$, suy ra r' và r thuộc cùng một thành phần liên thông mạnh, r' đã là chốt nên r không thể là chốt nữa.

Từ ba nhận xét và cách cài đặt chương trình như trong nhận xét 3, Ta có: Định r là chốt **nếu và chỉ nếu** không tồn tại cung ngược hoặc cung chéo nối một đỉnh thuộc nhánh DFS gốc r với một đỉnh ngoài nhánh đó, hay nói cách khác: **r là chốt nếu và chỉ nếu không tồn tại cung nối từ một đỉnh thuộc nhánh DFS gốc r tới một đỉnh thăm trước r** .

Dưới đây là một cài đặt hết sức thông minh, chỉ cần sửa đổi một chút thủ tục Visit ở trên là ta có ngay phương pháp này. Nội dung của nó là đánh số thứ tự các đỉnh từ đỉnh được thăm đầu tiên đến đỉnh thăm sau cùng. Định nghĩa Number[u] là số thứ tự của đỉnh u theo cách đánh số đó. Ta tính thêm Low[u] là giá trị Number[.] nhỏ nhất trong các đỉnh có thể đến được từ một đỉnh v nào đó của nhánh DFS gốc u bằng một cung (với giả thiết rằng u có một cung giả nối với chính u).

Cụ thể cách cực tiểu hoá Low[u] như sau:

Trong thủ tục Visit(u), trước hết ta đánh số thứ tự thăm cho đỉnh u và khởi gán Low[u] := Number[u] (u có cung tới chính u). Sau đó với mỗi đỉnh v nối từ u , có hai khả năng:

- ❖ Nếu v đã thăm thì ta cực tiểu hoá Low[u] theo công thức:

$$\text{Low}[u]_{\text{mới}} := \min(\text{Low}[u]_{\text{cũ}}, \text{Number}[v]).$$

- ❖ Nếu v chưa thăm thì ta gọi đệ quy đi thăm v , sau đó cực tiểu hoá Low[u] theo công thức:

$$\text{Low}[u]_{\text{mới}} := \min(\text{Low}[u]_{\text{cũ}}, \text{Low}[v])$$

Dễ dàng chứng minh được tính đúng đắn của công thức tính.

Khi duyệt xong một đỉnh u (chuẩn bị thoát khỏi thủ tục Visit(u)), ta so sánh Low[u] và Number[u]. Nếu như Low[u] = Number[u] thì u là chốt, bởi không có cung nối từ một đỉnh thuộc nhánh DFS gốc u tới một đỉnh thăm trước u . Khi đó chỉ việc liệt kê các đỉnh thuộc thành phần liên thông mạnh chứa u là nhánh DFS gốc u .

Để công việc dễ dàng hơn nữa, ta định nghĩa một danh sách Stack được tổ chức dưới dạng ngăn xếp và dùng ngăn xếp này để lấy ra các đỉnh thuộc một nhánh nào đó. Khi thăm tới một đỉnh u , ta đẩy ngay đỉnh u đó vào ngăn xếp, thì khi duyệt xong đỉnh u , mọi đỉnh thuộc nhánh DFS gốc u sẽ được đẩy vào ngăn xếp Stack ngay sau u . Nếu u là chốt, ta chỉ việc lấy các đỉnh ra khỏi ngăn xếp Stack cho tới khi lấy tới đỉnh u là sẽ được nhánh DFS gốc u cũng chính là thành phần liên thông mạnh chứa u .

Thuật toán Tarjan:

```

procedure Visit(uεV);
begin
    Count := Count + 1; Number[u] := Count; {Trước hết đánh số u}
    Low[u] := Number[u];
    Push(u); {Đẩy u vào ngăn xếp}
    {Đánh dấu u đã thăm};
    for (forall v: (u, v)εE) do
        if {v đã thăm} then
            Low[u] := min(Low[u], Number[v])
        else
            begin
                Visit(v);
                Low[u] := min(Low[u], Low[v]);
            end;
    if Number[u] = Low[u] then {Nếu u là chốt}
        begin
            {Thông báo thành phần liên thông mạnh với chốt u gồm có các đỉnh:};
            repeat
                v := Pop; {Lấy từ ngăn xếp ra một đỉnh v}
                {Output v};
                {Xoá đỉnh v khỏi đồ thị};
            until v = u;
        end;
    end;

begin
    {Thêm vào đồ thị một đỉnh x và các cung (x, v) với mọi v};
    Count := 0;
    L := Ø; {Khởi tạo một ngăn xếp rỗng}
    Visit(x)
end.

```

Bởi thuật toán Tarjan chỉ là sửa đổi một chút thuật toán DFS, các thao tác vào/ra ngăn xếp được thực hiện không quá n lần. Vậy nên nếu đồ thị có n đỉnh và m cung thì độ phức tạp tính toán của thuật toán Tarjan vẫn là $O(n + m)$ trong trường hợp biểu diễn đồ thị bằng danh sách kè, là $O(n^2)$ trong trường hợp biểu diễn bằng ma trận kè và là $O(n.m)$ trong trường hợp biểu diễn bằng danh sách cạnh.


Mọi thứ đã sẵn sàng, dưới đây là toàn bộ chương trình. Trong chương trình này, ta sử dụng:

- ❖ Ma trận kè A để biểu diễn đồ thị.
- ❖ Mảng Free kiểu Boolean, Free[u] = True nếu u chưa bị liệt kê vào thành phần liên thông nào, tức là u chưa bị loại khỏi đồ thị.
- ❖ Mảng Number và Low với công dụng như trên, quy ước Number[u] = 0 nếu đỉnh u chưa được thăm.
- ❖ Mảng Stack, thủ tục Push, hàm Pop để mô tả cấu trúc ngăn xếp.

Input: file văn bản SCONNECT.INP:

- ❖ Dòng đầu: Ghi số đỉnh n (≤ 100) và số cung m của đồ thị cách nhau một dấu cách
- ❖ m dòng tiếp theo, mỗi dòng ghi hai số nguyên u, v cách nhau một dấu cách thể hiện có cung (u, v) trong đồ thị

Output: file văn bản SCONNECT.OUT, liệt kê các thành phần liên thông mạnh



SCONNECT.INP	SCONNECT.OUT
11 15	Component 1:
1 2	7, 6, 5,
1 8	Component 2:
2 3	4, 3, 2,
3 4	Component 3:
4 2	11, 10, 9, 8,
4 5	Component 4:
5 6	1,
6 7	
7 5	
8 9	
9 4	
9 10	
10 8	
10 11	
11 8	

P_4_04_2.PAS * Thuật toán Tarjan liệt kê các thành phần liên thông mạnh

{\$MODE DELPHI} (*This program uses 32-bit Integer [-2³¹..2³¹ - 1]*)

```
program Strong_connectivity;
const
  InputFile = 'SCONNECT.INP';
  OutputFile = 'SCONNECT.OUT';
  max = 100;
var
  a: array[1..max, 1..max] of Boolean;
  Free: array[1..max] of Boolean;
  Number, Low, Stack: array[1..max] of Integer;
  n, Count, ComponentCount, Top: Integer;
  fo: Text;

procedure Enter;
var
  i, u, v, m: Integer;
  fi: Text;
begin
  Assign(fi, InputFile); Reset(fi);
  FillChar(a, SizeOf(a), False);
  ReadLn(fi, n, m);
  for i := 1 to m do
    begin
      ReadLn(fi, u, v);
      a[u, v] := True;
    end;
  Close(fi);
end;

procedure Init; {Khởi tạo}
begin
  FillChar(Number, SizeOf(Number), 0); {Mọi đỉnh đều chưa thăm}
  FillChar(Free, SizeOf(Free), True); {Chưa đỉnh nào bị loại}
  Top := 0; {Ngăn xếp rỗng}
  Count := 0; {Biến đánh số thứ tự thăm}
  ComponentCount := 0; {Biến đánh số các thành phần liên thông}
end;

procedure Push(v: Integer); {Đẩy một đỉnh v vào ngăn xếp}
begin
  Inc(Top);
  Stack[Top] := v;
end;
```

```

function Pop: Integer; {Lấy một đỉnh khỏi ngăn xếp, trả về trong kết quả hàm}
begin
  Pop := Stack[Top];
  Dec(Top);
end;

function Min(x, y: Integer): Integer;
begin
  if x < y then Min := x else Min := y;
end;

procedure Visit(u: Integer); {Thuật toán tìm kiếm theo chiều sâu bắt đầu từ u}
var
  v: Integer;
begin
  Inc(Count); Number[u] := Count; {Trước hết đánh số cho u}
  Low[u] := Number[u]; {Coi u có cung tới u, nên có thể khởi gán Low[u] thê này rồi sau cực tiêu hoá dần}
  Push(u); {Đẩy u vào ngăn xếp}
  for v := 1 to n do
    if Free[v] and a[u, v] then {Xét những đỉnh v kề u}
      if Number[v] <> 0 then {Nếu v đã thăm}
        Low[u] := Min(Low[u], Number[v]); {Cực tiêu hoá Low[u] theo công thức này}
      else {Nếu v chưa thăm}
        begin
          Visit(v); {Tiếp tục tìm kiếm theo chiều sâu bắt đầu từ v}
          Low[u] := Min(Low[u], Low[v]); {Rồi cực tiêu hoá Low[u] theo công thức này}
        end;
    {Đến đây thì đỉnh u được duyệt xong, tức là các đỉnh thuộc nhánh DFS gốc u đều đã thăm}
    if Number[u] = Low[u] then {Nếu u là chốt}
      begin {Liệt kê thành phần liên thông mạnh có chốt u}
        Inc(ComponentCount);
        WriteLn(fo, 'Component ', ComponentCount, ': ');
        repeat
          v := Pop; {Lấy dần các đỉnh ra khỏi ngăn xếp}
          Write(fo, v, ', ');
          Free[v] := False; {Rồi loại luôn khỏi đồ thị}
        until v = u; {Cho tới khi lấy tới đỉnh u}
        WriteLn(fo);
      end;
    end;
end;

procedure Solve;
var
  u: Integer;
begin
  {Thay vì thêm một đỉnh giả x và các cung (x, v) với mọi đỉnh v rồi gọi Visit(x), ta có thể làm thê này cho nhanh}
  for u := 1 to n do
    if Number[u] = 0 then Visit(u);
end;

begin
  Enter;
  Assign(fo, OutputFile); Rewrite(fo);
  Init;
  Solve;
  Close(fo);
end.

```

Bài tập

Bài 1

Phương pháp cài đặt như trên có thể nói là rất hay và hiệu quả, đòi hỏi ta phải hiểu rõ bản chất thuật toán, nếu không thì rất dễ nhầm. Trên thực tế, còn có một phương pháp khác dễ hiểu hơn, tuy tính hiệu quả có kém hơn một chút. Hãy viết chương trình mô tả phương pháp sau:

Vẫn dùng thuật toán tìm kiếm theo chiều sâu với thủ tục Visit nói ở đầu mục, đánh số lại các đỉnh từ 1 tới n theo thứ tự **duyệt xong**, sau đó đảo chiều tất cả các cung của đồ thị. Xét lần lượt các đỉnh theo thứ tự từ đỉnh duyệt xong sau cùng tới đỉnh duyệt xong đầu tiên, với mỗi đỉnh đó, ta lại dùng thuật toán tìm kiếm trên đồ thị (BFS hay DFS) liệt kê những đỉnh nào đến được từ đỉnh đang xét, đó chính là một thành phần liên thông mạnh. Lưu ý là khi liệt kê xong thành phần nào, ta loại ngay các đỉnh của thành phần đó khỏi đồ thị.


Tính đúng đắn của phương pháp có thể hình dung không mấy khó khăn:

Trước hết ta thêm vào đồ thị đỉnh x và các cung (x, v) với mọi v, sau đó gọi Visit(x) để xây dựng cây DFS gốc x. Hiển nhiên x là chốt của thành phần liên thông chỉ gồm mỗi x. Sau đó bỏ đỉnh x khỏi cây DFS, cây sẽ phân rã thành các cây con.

Đỉnh r duyệt xong sau cùng chắc chắn là gốc của một cây con (bởi khi duyệt xong nó chắc chắn sẽ lùi về x) suy ra r là chốt. Hơn thế nữa, nếu một đỉnh u nào đó tới được r thì u cũng phải thuộc cây con gốc r. Bởi nếu giả sử phản chứng rằng u thuộc cây con khác thì u phải được thăm trước r (do cây con gốc r được thăm tới sau cùng), có nghĩa là khi Visit(u) thì r chưa thăm. Vậy nên r sẽ thuộc nhánh DFS gốc u, mâu thuẫn với lập luận r là gốc. Từ đó suy ra nếu u tới được r thì r tới được u, tức là khi đảo chiều các cung, nếu r tới được đỉnh nào thì đỉnh đó thuộc thành phần liên thông chốt r.

Loại bỏ thành phần liên thông với chốt r khỏi đồ thị. Cây con gốc r lại phân rã thành nhiều cây con. Lập luận tương tự như trên với v' là đỉnh duyệt xong sau cùng (Hình 66).

Ví dụ:



Hình 66: Đánh số lại, đảo chiều các cung và duyệt BFS với cách chọn các đỉnh xuất phát ngược lại với thứ tự duyệt xong (thứ tự 11, 10... 3, 2, 1)

Bài 2

Thuật toán Warshall có thể áp dụng tìm bao đóng của đồ thị có hướng, vậy hãy kiểm tra tính liên thông mạnh của một đồ thị có hướng bằng hai cách: Dùng các thuật toán tìm kiếm trên đồ thị và thuật toán Warshall, sau đó so sánh ưu, nhược điểm của mỗi phương pháp

Bài 3

Trên mặt phẳng với hệ tọa độ Decartes vuông góc cho n đường tròn, mỗi đường tròn xác định bởi bộ 3 số thực (X, Y, R) ở đây (X, Y) là tọa độ tâm và R là bán kính. Hai đường tròn gọi là thông nhau nếu chúng có điểm chung. Hãy chia các đường tròn thành một số tối thiểu các nhóm sao cho hai đường tròn bất kỳ trong một nhóm bất kỳ có thể đi được sang nhau sau một số hữu hạn các bước di chuyển giữa hai đường tròn thông nhau.

§5. VÀI ỨNG DỤNG CỦA DFS VÀ BFS

5.1. XÂY DỰNG CÂY KHUNG CỦA ĐỒ THỊ

Cây là đồ thị **vô hướng, liên thông, không có chu trình đơn**. Đồ thị vô hướng không có chu trình đơn gọi là rừng (hợp của nhiều cây). Như vậy mỗi thành phần liên thông của rừng là một cây.

Khái niệm cây được sử dụng rộng rãi trong nhiều lĩnh vực khác nhau: Nghiên cứu cấu trúc các phân tử hữu cơ, xây dựng các thuật toán tổ chức thư mục, các thuật toán tìm kiếm, lưu trữ và nén dữ liệu...

5.1.1. Định lý (Daisy Chain Theorem)

Giả sử $T = (V, E)$ là đồ thị vô hướng với n đỉnh. Khi đó các mệnh đề sau là tương đương:

1. T là cây
2. T không chứa chu trình đơn và có $n - 1$ cạnh
3. T liên thông và mỗi cạnh của nó đều là cầu
4. Giữa hai đỉnh bất kỳ của T đều tồn tại đúng một đường đi đơn
5. T không chứa chu trình đơn nhưng nếu thêm vào **một cạnh** ta thu được một chu trình đơn.
6. T liên thông và có $n - 1$ cạnh

Chứng minh:

$1 \Rightarrow 2$: “**T là cây**” \Rightarrow “**T không chứa chu trình đơn và có $n - 1$ cạnh**”

Từ T là cây, theo định nghĩa T không chứa chu trình đơn. Ta sẽ chứng minh cây T có n đỉnh thì phải có $n - 1$ cạnh bằng quy nạp theo số đỉnh n . Rõ ràng khi $n = 1$ thì cây có 1 đỉnh sẽ không có cạnh. Nếu $n > 1$ thì do đồ thị hữu hạn nên số các đường đi đơn trong T cũng hữu hạn, gọi $P = (v_1, v_2, \dots, v_k)$ là một đường đi dài nhất (qua nhiều cạnh nhất) trong T . Đỉnh v_1 không thể có cạnh nối với đỉnh nào trong số các đỉnh v_3, v_4, \dots, v_k . Bởi nếu có cạnh (v_1, v_p) ($3 \leq p \leq k$) thì ta sẽ thiết lập được chu trình đơn $(v_1, v_2, \dots, v_p, v_1)$. Mặt khác, đỉnh v_1 cũng không thể có cạnh nối với đỉnh nào khác ngoài các đỉnh trên P bởi nếu có cạnh (v_1, v_0) ($v_0 \notin P$) thì ta thiết lập được đường đi $(v_0, v_1, v_2, \dots, v_k)$ dài hơn đường đi P . Vậy đỉnh v_1 chỉ có đúng một cạnh nối với v_2 hay v_1 là đỉnh treo. Loại bỏ v_1 và cạnh (v_1, v_2) khỏi T ta được đồ thị mới cũng là cây và có $n - 1$ đỉnh, cây này theo giả thiết quy nạp có $n - 2$ cạnh. Vậy cây T có $n - 1$ cạnh.

$2 \Rightarrow 3$: “**T không chứa chu trình đơn và có $n - 1$ cạnh**” \Rightarrow “**T liên thông và mỗi cạnh của nó đều là cầu**”

Giả sử T có k thành phần liên thông T_1, T_2, \dots, T_k . Vì T không chứa chu trình đơn nên các thành phần liên thông của T cũng không chứa chu trình đơn, tức là các T_1, T_2, \dots, T_k đều là cây. Gọi n_1, n_2, \dots, n_k lần lượt là số đỉnh của T_1, T_2, \dots, T_k thì cây T_1 có $n_1 - 1$ cạnh, cây T_2 có $n_2 - 1$ cạnh..., cây T_k có $n_k - 1$ cạnh. Cộng lại ta có số cạnh của T là $n_1 + n_2 + \dots + n_k - k = n - 1$.

k cạnh. Theo giả thiết, cây T có $n - 1$ cạnh, suy ra $k = 1$, đồ thị chỉ có một thành phần liên thông là đồ thị liên thông.

Bây giờ khi T đã liên thông, nếu bỏ đi một cạnh của T thì T sẽ còn $n - 2$ cạnh và sẽ không liên thông bởi nếu T vẫn liên thông thì do T không có chu trình nên T sẽ là cây và có $n - 1$ cạnh. Điều đó chứng tỏ mỗi cạnh của T đều là cầu.

3⇒4: “T liên thông và mỗi cạnh của nó đều là cầu” ⇒ “Giữa hai đỉnh bất kỳ của T có đúng một đường đi đơn”

Gọi x và y là 2 đỉnh bất kỳ trong T, vì T liên thông nên sẽ có một đường đi đơn từ x tới y. Nếu tồn tại một đường đi đơn khác từ x tới y thì nếu ta bỏ đi một cạnh (u, v) nằm trên đường đi thứ nhất nhưng không nằm trên đường đi thứ hai thì từ u vẫn có thể đến được v bằng cách: đi từ u đi theo chiều tới x theo các cạnh thuộc đường thứ nhất, sau đó đi từ x tới y theo đường thứ hai, rồi lại đi từ y tới v theo các cạnh thuộc đường đi thứ nhất. Điều này mâu thuẫn với giả thiết (u, v) là cầu.

4⇒5: “Giữa hai đỉnh bất kỳ của T có đúng một đường đi đơn” ⇒ “T không chứa chu trình đơn nhưng sẽ có thêm vào một cạnh ta thu được một chu trình đơn”

Thứ nhất T không chứa chu trình đơn vì nếu T chứa chu trình đơn thì chu trình đó qua ít nhất hai đỉnh u, v . Rõ ràng dọc theo các cạnh trên chu trình đó thì từ u có hai đường đi đơn tới v . Vô lý.

Giữa hai đỉnh u, v bất kỳ của T có một đường đi đơn nối u với v , vậy khi thêm cạnh (u, v) vào đường đi này thì sẽ tạo thành chu trình.

5⇒6: “T không chứa chu trình đơn nhưng sẽ có thêm vào một cạnh ta thu được một chu trình đơn” ⇒ “T liên thông và có $n - 1$ cạnh”

Gọi u và v là hai đỉnh bất kỳ trong T, thêm vào T một cạnh (u, v) nữa thì theo giả thiết sẽ tạo thành một chu trình chứa cạnh (u, v). Loại bỏ cạnh này đi thì phần còn lại của chu trình sẽ là một đường đi từ u tới v . Mọi cặp đỉnh của T đều có một đường đi nối chúng tức là T liên thông, theo giả thiết T không chứa chu trình đơn nên T là cây và có $n - 1$ cạnh.


6⇒1: “T liên thông và có $n - 1$ cạnh” ⇒ “T là cây”

Giả sử T không là cây thì T có chu trình, huỷ bỏ một cạnh trên chu trình này thì T vẫn liên thông, nếu đồ thị mới nhận được vẫn có chu trình thì lại huỷ một cạnh trong chu trình mới. Cứ như thế cho tới khi ta nhận được một đồ thị liên thông không có chu trình. Đồ thị này là cây nhưng lại có $< n - 1$ cạnh (vô lý). Vậy T là cây

5.1.2. Định nghĩa

Giả sử $G = (V, E)$ là đồ thị vô hướng. Cây $T = (V, F)$ với $F \subseteq E$ gọi là cây khung của đồ thị G. Tức là nếu như loại bỏ một số cạnh của G để được một cây thì cây đó gọi là cây khung (hay cây bao trùm của đồ thị).

Dễ thấy rằng với một đồ thị vô hướng liên thông có thể có nhiều cây khung (Hình 67).



Hình 67: Đồ thị \$G\$ và một số ví dụ cây khung \$T_1, T_2, T_3\$ của nó

Điều kiện cần và đủ để một đồ thị vô hướng có cây khung là đồ thị đó phải liên thông

Số cây khung của đồ thị đầy đủ \$K_n\$ là \$n^{n-2}\$.

5.1.3. Thuật toán xây dựng cây khung


Xét đồ thị vô hướng liên thông \$G = (V, E)\$ có \$n\$ đỉnh, có nhiều thuật toán xây dựng cây khung của \$G\$

Xây dựng cây khung bằng thuật toán hợp nhất

Trước hết, đặt \$T = (V, \emptyset)\$; \$T\$ không chứa cạnh nào thì có thể coi \$T\$ gồm \$n\$ cây rác, mỗi cây chỉ có 1 đỉnh. Sau đó xét lần lượt các cạnh của \$G\$, nếu cạnh đang xét nối hai cây khác nhau trong \$T\$ thì thêm cạnh đó vào \$T\$, đồng thời hợp nhất hai cây đó lại thành một cây. Cứ làm như vậy cho tới khi kết nạp đủ \$n - 1\$ cạnh vào \$T\$ thì ta được \$T\$ là cây khung của đồ thị. Các phương pháp kiểm tra cạnh có nối hai cây khác nhau hay không cũng như kỹ thuật hợp nhất hai cây sẽ được bàn kỹ hơn trong thuật toán Kruskal ở [§9 - mục 9.2].

Xây dựng cây khung bằng các thuật toán tìm kiếm trên đồ thị.

Áp dụng thuật toán BFS hay DFS bắt đầu từ đỉnh \$S\$, tại mỗi bước từ đỉnh \$u\$ tới thăm đỉnh \$v\$, ta thêm vào thao tác ghi nhận luôn cạnh \$(u, v)\$ vào cây khung. Do đồ thị liên thông nên thuật toán sẽ xuất phát từ \$S\$ và tới thăm tất cả các đỉnh còn lại, mỗi đỉnh đúng một lần, tức là quá trình duyệt sẽ ghi nhận được đúng \$n - 1\$ cạnh. Tất cả những cạnh đó không tạo thành chu trình đơn bởi thuật toán không thăm lại những đỉnh đã thăm. Theo mệnh đề tương đương thứ hai, ta có những cạnh ghi nhận được tạo thành một cây khung của đồ thị.



Hình 68: Cây khung DFS (a) và cây khung BFS (b) (Mũi tên chỉ chiều đi thăm các đỉnh)

5.2. TẬP CÁC CHU TRÌNH CƠ SỞ CỦA ĐỒ THỊ

Xét một đồ thị vô hướng liên thông $G = (V, E)$; gọi $T = (V, F)$ là một cây khung của nó. Các cạnh của cây khung được gọi là các cạnh trong, còn các cạnh khác là các cạnh ngoài.

Nếu thêm một cạnh ngoài $e \in E \setminus F$ vào cây khung T , thì ta được đúng một chu trình đơn trong T , ký hiệu chu trình này là C_e . Tập các chu trình:

$$\Psi = \{C_e \mid e \in E \setminus F\}$$

được gọi là tập các chu trình cơ sở của đồ thị G .

Các tính chất quan trọng của tập các chu trình cơ sở:

- ❖ Tập các chu trình cơ sở là phụ thuộc vào cây khung, hai cây khung khác nhau có thể cho hai tập chu trình cơ sở khác nhau.
- ❖ Nếu đồ thị liên thông có n đỉnh và m cạnh, thì trong cây khung có $n - 1$ cạnh, còn lại $m - n + 1$ cạnh ngoài. Tương ứng với mỗi cạnh ngoài có một chu trình cơ sở, vậy **số chu trình cơ sở của đồ thị liên thông là $m - n + 1$** .
- ❖ **Tập các chu trình cơ sở là tập nhiều nhất các chu trình thoả mãn:** Mỗi chu trình có đúng một cạnh riêng, cạnh đó không nằm trong bất cứ một chu trình nào khác. Bởi nếu có một tập gồm t chu trình thoả mãn điều đó thì việc loại bỏ cạnh riêng của một chu trình sẽ không làm mất tính liên thông của đồ thị, đồng thời không ảnh hưởng tới sự tồn tại của các chu trình khác. Như vậy nếu loại bỏ tất cả các cạnh riêng thì đồ thị vẫn liên thông và còn $m - t$ cạnh. Đồ thị liên thông thì không thể có ít hơn $n - 1$ cạnh nên ta có $m - t \geq n - 1$ hay $t \leq m - n + 1$.
- ❖ **Mọi cạnh trong một chu trình đơn bất kỳ đều phải thuộc một chu trình cơ sở:** Bởi nếu có một cạnh (u, v) không thuộc một chu trình cơ sở nào, thì khi ta bỏ cạnh đó đi đồ thị vẫn liên thông và không ảnh hưởng tới sự tồn tại của các chu trình cơ sở. Lại bỏ tiếp những cạnh ngoài của các chu trình cơ sở thì đồ thị vẫn liên thông và còn lại $m - (m - n + 1) - 1 = n - 2$ cạnh. Điều này vô lý.

Đối với đồ thị $G = (V, E)$ có n đỉnh và m cạnh, có k thành phần liên thông, ta có thể xét các thành phần liên thông và xét riêng các cây khung của các thành phần đó. Khi đó có thể mở rộng khái niệm tập các chu trình cơ sở cho đồ thị vô hướng tổng quát: Mỗi khi thêm một cạnh không nằm trong các cây khung vào rừng, ta được đúng một chu trình đơn, tập các chu trình đơn tạo thành bằng cách ghép các cạnh ngoài như vậy gọi là tập các chu trình cơ sở của đồ thị G . **Số các chu trình cơ sở là $m - n + k$.**

5.3. BÀI TOÁN ĐỊNH CHIỀU ĐỒ THỊ

Bài toán đặt ra là cho một đồ thị vô hướng liên thông $G = (V, E)$, hãy thay mỗi cạnh của đồ thị bằng một cung định hướng để được một đồ thị có hướng liên thông mạnh. Nếu có phương án định chiều như vậy thì G được gọi là đồ thị định chiều được. Bài toán định chiều đồ thị có ứng dụng rõ nhất trong sơ đồ giao thông đường bộ. Chẳng hạn như trả lời câu hỏi: Trong một

hệ thống đường phố, liệu có thể quy định các đường phố đó thành đường một chiều mà vẫn đảm bảo sự đi lại giữa hai nút giao thông bất kỳ hay không.

Có thể tổng quát hoá bài toán định chiều đồ thị: Với đồ thị vô hướng $G = (V, E)$ hãy tìm cách thay mỗi cạnh của đồ thị bằng một cung định hướng để được đồ thị mới có ít thành phần liên thông mạnh nhất. Dưới đây ta xét một tính chất hữu ích của thuật toán thuật toán tìm kiếm theo chiều sâu để giải quyết bài toán định chiều đồ thị

5.3.1. Phép định chiều DFS

Xét mô hình duyệt đồ thị bằng thuật toán tìm kiếm theo chiều sâu:

```
procedure Visit(u ∈ V);
begin
    ⟨Thông báo thăm u và đánh dấu u đã thăm⟩;
    for (∀v: (u, v) ∈ E) do
        if ⟨v chưa thăm⟩ then Visit(v);
    end;

    begin
        ⟨Đánh dấu mọi đỉnh đều chưa thăm⟩;
        for (∀v ∈ V) do
            if ⟨v chưa thăm⟩ then Visit(v);
    end;
```

Coi một cạnh của đồ thị tương đương với hai cung có hướng ngược chiều nhau. Thuật toán tìm kiếm theo chiều sâu theo mô hình trên sẽ duyệt qua hết các đỉnh của đồ thị và tất cả các cung nữa. Quá trình duyệt cho ta một cây tìm kiếm DFS. Ta có các nhận xét sau:


Nhận xét 1:

Quá trình duyệt sẽ không có cung chéo (cung đi từ một nhánh DFS thăm sau tới nhánh DFS thăm trước). Thật vậy, nếu quá trình duyệt xét tới một cung (u, v) :

- ❖ Nếu u thăm trước v có nghĩa là khi $\text{Visit}(u)$ được gọi thì v chưa thăm, vì thủ tục $\text{Visit}(u)$ sẽ xây dựng nhánh DFS gốc u gồm những đỉnh chưa thăm đến được từ u , suy ra v nằm trong nhánh DFS gốc $u \Rightarrow v$ là hậu duệ của u , hay (u, v) là cung DFS hoặc cung xuôi.
- ❖ Nếu u thăm sau v (v thăm trước u), tương tự trên, ta suy ra u nằm trong nhánh DFS gốc v , v là tiền bối của $u \Rightarrow (u, v)$ là cung ngược.

Nhận xét 2:

Trong quá trình duyệt đồ thị theo chiều sâu, nếu cứ duyệt qua cung (u, v) nào thì ta bỏ đi cung (v, u) . (Tức là sẽ duyệt qua cung (u, v) thì ta định chiều luôn cạnh (u, v) theo chiều từ u tới v), ta được một phép định chiều đồ thị gọi là phép định chiều DFS.



Hình 69: Phép định chiều DFS

Nhận xét 3:

Với phép định chiều DFS, thì sẽ chỉ còn các cung trên cây DFS và cung ngược, không còn lại cung xuôi. Bởi trên đồ thị vô hướng ban đầu, nếu ta coi một cạnh là hai cung có hướng ngược chiều nhau thì với một cung xuôi ta có cung ngược chiều với nó là cung ngược. Do tính chất DFS, cung ngược được duyệt trước cung xuôi tương ứng, nên khi định chiều cạnh theo cung ngược thì cung xuôi sẽ bị huỷ và không bị xét tới nữa.

Nhận xét 4:

Trong đồ thị vô hướng ban đầu, cạnh bị định hướng thành cung ngược chính là cạnh ngoài của cây DFS. Chính vì vậy, **mọi chu trình cơ sở của cây DFS trong đồ thị vô hướng ban đầu vẫn sẽ là chu trình** trong đồ thị có hướng tạo ra. (Đây là một phương pháp hiệu quả để liệt kê các chu trình cơ sở của cây khung DFS: Vừa duyệt DFS vừa định chiều, nếu duyệt phải cung ngược (u, v) thì truy vết đường đi của DFS để tìm đường từ v đến u , sau đó nối thêm cung ngược (u, v) để được một chu trình cơ sở).

Định lý: Điều kiện cần và đủ để một đồ thị vô hướng liên thông có thể định chiều được là mỗi cạnh của đồ thị nằm trên ít nhất một chu trình đơn (Hay nói cách khác mọi cạnh của đồ thị đều không phải là cầu).

Chứng minh: Gọi $G = (V, E)$ là một đồ thị vô hướng liên thông.

" \Rightarrow "

Nếu G là định chiều được thì sau khi định hướng sẽ được đồ thị liên thông mạnh G' . Với một cạnh (u, v) được định chiều thành cung (u, v) thì sẽ tồn tại một đường đi đơn trong G' theo các cạnh định hướng từ v về u . Đường đi đó nối thêm cung (u, v) sẽ thành một chu trình đơn có hướng trong G' . Tức là trên đồ thị ban đầu, cạnh (u, v) nằm trên một chu trình đơn.

" \Leftarrow "

Nếu mỗi cạnh của G đều nằm trên một chu trình đơn, ta sẽ chứng minh rằng: phép định chiều DFS sẽ tạo ra đồ thị G' liên thông mạnh.

Trước hết ta chứng minh rằng nếu (u, v) là cạnh của G thì sẽ có một đường đi từ u tới v trong G' . Thật vậy, vì (u, v) nằm trong một chu trình đơn, mà mọi cạnh của một chu trình đơn đều

phải thuộc một chu trình cơ sở của cây DFS, nên sẽ có một chu trình cơ sở chứa cả u và v . Chu trình cơ sở của cây DFS qua phép định chiều DFS vẫn là chu trình trong G' nên đi theo các cạnh định hướng của chu trình đó, ta có thể đi từ u tới v và ngược lại trên G' .

Nếu u và v là 2 đỉnh bất kỳ của G thì do G liên thông, tồn tại một đường đi ($u=x[0], x[1], \dots, x[n]=v$). Vì $(x[i], x[i+1])$ là cạnh của G nên trong G' , từ $x[i]$ có thể đến được $x[i+1]$. Suy ra từ u cũng có thể đến được v bằng các cạnh định hướng của G' .


Với những kết quả đã chứng minh trên, ta còn suy ra được: Nếu đồ thị liên thông và mỗi cạnh của nó nằm trên ít nhất một chu trình đơn thì phép định chiều DFS sẽ cho một đồ thị liên thông mạnh. Còn nếu không, thì phép định chiều DFS sẽ cho một đồ thị định hướng có ít thành phần liên thông mạnh nhất, một cạnh không nằm trên một chu trình đơn nào (cầu) của đồ thị ban đầu sẽ được định hướng thành cung nối giữa hai thành phần liên thông mạnh.

Giải thuật cho bài toán định chiều đồ thị đến đây quá đơn giản nên có lẽ không cần cài đặt cụ thể chương trình nữa. Bây giờ song song với việc định chiều, ta mô tả phương pháp đánh số và ghi nhận cung ngược lên trên cao nhất từ một nhánh cây DFS để dùng vào một số bài toán khác.

5.3.2. Phép đánh số và ghi nhận cung ngược lên cao nhất trên cây DFS

Trong quá trình định chiều, ta thêm vào đó thao tác đánh số các đỉnh theo thứ tự thăm DFS, gọi $\text{Number}[u]$ là số thứ tự của đỉnh u theo cách đánh số đó. Định nghĩa thêm $\text{Low}[u]$ là giá trị $\text{Number}[\cdot]$ nhỏ nhất của những đỉnh đến được từ nhánh DFS gốc u bằng một cung ngược. Tức là nếu nhánh DFS gốc u có nhiều cung ngược hướng lên phía gốc thì ta ghi nhận lại cung ngược hướng lên cao nhất. Nếu nhánh DFS gốc u không chứa cung ngược thì ta cho $\text{Low}[u] = +\infty$. Cụ thể cách cực tiểu hóa $\text{Low}[u]$ như sau: Trong thủ tục $\text{Visit}(u)$, trước hết ta đánh số thứ tự thăm cho đỉnh u ($\text{Number}[u]$) và khởi gán $\text{Low}[u] = +\infty$, sau đó xét tất cả những đỉnh v kề u , định chiều cạnh (u, v) thành cung (u, v) . Có hai khả năng xảy ra:

- ❖ v chưa thăm thì ta gọi $\text{Visit}(v)$ để thăm v , khi thủ tục $\text{Visit}(v)$ thoát có nghĩa là đã xây dựng được nhánh DFS gốc $v \subset$ nhánh DFS gốc u , khi đó những cung ngược đi từ nhánh DFS gốc v có thể coi là cung ngược đi từ nhánh DFS gốc $u \Rightarrow$ ta cực tiểu hóa $\text{Low}[u]$ theo công thức: $\text{Low}[u]_{\text{mới}} := \min(\text{Low}[u]_{\text{cũ}}, \text{Low}[v])$
- ❖ v đã thăm thì (u, v) là một cung ngược đi từ nhánh DFS gốc $u \Rightarrow$ ta sẽ cực tiểu hóa $\text{Low}[u]$ theo công thức: $\text{Low}[u]_{\text{mới}} := \min(\text{Low}[u]_{\text{cũ}}, \text{Number}[v])$



Hình 70: Phép đánh số và ghi nhận cung ngược lên cao nhất

5.4. LIỆT KÊ CÁC KHỚP VÀ CẦU CỦA ĐỒ THI

Với một đồ thị vô hướng $G = (V, E)$, ta dùng phép định chiều DFS để định chiều G , khi đó hãy để ý một cung (u, v) là cung trên cây DFS, ta có các nhận xét sau:

- ❖ Nếu từ nhánh DFS gốc v không có cung nào ngược lên phía trên v có nghĩa là từ một đỉnh thuộc nhánh DFS gốc v đi theo các cung định hướng chỉ đi được tới những đỉnh nội bộ trong nhánh DFS gốc v mà thôi, suy ra (u, v) là một cầu. Cũng dễ dàng chứng minh được điều ngược lại. Vậy (u, v) là cầu $\Leftrightarrow \text{Low}[v] \geq \text{Number}[v]$.
 - ❖ Nếu từ nhánh DFS gốc v không có cung nào ngược lên phía trên u, tức là nếu bỏ u đi thì từ v không có cách nào lên được các tiền bối của u. Điều này chỉ ra rằng nếu u không phải là nút gốc của một cây DFS thì u là khốp. Cũng dễ dàng chứng minh điều ngược lại. Vậy nếu u không là gốc của một cây DFS thì u là khốp $\Leftrightarrow \text{Low}[v] \geq \text{Number}[u]$.
 - ❖ Nếu r là gốc của một cây DFS thì r là khốp $\Leftrightarrow r$ có ít nhất 2 nhánh con


Đến đây ta đã có đủ điều kiện để giải bài toán liệt kê các khốp và cầu của đồ thị: đơn giản là dùng phép định chiều DFS đánh số các đỉnh theo thứ tự thăm và ghi nhận cung ngược lên trên cao nhất xuất phát từ một nhánh cây DFS, sau đó dùng ba nhận xét kể trên để liệt kê ra tất cả các cầu và khốp của đồ thị:

Input: file văn bản CUT.INP, trong đó

- ❖ Dòng 1: Chứa số đỉnh n ($n \leq 100$) và số cạnh m của đơn đồ thị vô hướng G cách nhau ít nhất một dấu cách
 - ❖ m dòng tiếp theo, mỗi dòng ghi hai số u, v cách nhau ít nhất một dấu cách, thể hiện (u, v) là một cạnh của G

Output: file văn bản CUT.OUT, trong đó ghi:

- ❖ Danh sách các câu
 - ❖ Danh sách các khớp



Ngoài các mảng đã được nói tới khi trình bày thuật toán, có thêm một mảng Parent[1..n], trong đó Parent[v] chỉ ra nút cha của nút v trên cây DFS, nếu v là gốc của một cây DFS thì Parent[v] được đặt bằng -1. Công dụng của mảng Parent là cho phép duyệt tất cả các cạnh trên cây DFS và kiểm tra một đỉnh có phải là gốc của cây DFS hay không.

P_4_05_1.PAS * Liệt kê các khốp và cầu của đồ thị

{\$MODE DELPHI} (*This program uses 32-bit Integer [-2³¹..2³¹ - 1]*)

```
program Directivity_Bridges_CutVertices;
const
  InputFile = 'CUT.INP';
  OutputFile = 'CUT.OUT';
  max = 100;
var
  a: array[1..max, 1..max] of Boolean;
  Number, Low, Parent: array[1..max] of Integer;
  n, Count: Integer;

procedure Enter; {Nhập dữ liệu}
var
  f: Text;
  i, m, u, v: Integer;
begin
  FillChar(a, SizeOf(a), False);
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n, m);
  for i := 1 to m do
    begin
      ReadLn(f, u, v);
      a[u, v] := True;
      a[v, u] := True;
    end;
  Close(f);
end;

procedure Visit(u: Integer); {Duyệt DFS, định chiều, đánh số}
var
  v: Integer;
begin
  Inc(Count); Number[u] := Count; {Trước hết đánh số u}
  Low[u] := n + 1; {Khởi điểm Low[u] = +∞}
  for v := 1 to n do
    if a[u, v] then {Xét ∀v kề u}
      begin
        a[v, u] := False; {Bỏ đi cung (v, u)}
        Visit(v);
      end;
end;
```

```

if Parent[v] = 0 then {Parent[v] = 0 ⇔ v chưa thăm, (u, v) là cung DFS}
begin
  Parent[v] := u;
  Visit(v); {Thăm v}
  if Low[u] > Low[v] then Low[u] := Low[v]; {Cực tiêu hoá Low[u] theo Low[v]}
end
else {v đã thăm, (u, v) là cung ngược}
  if Low[u] > Number[v] then Low[u] := Number[v]; {Cực tiêu hoá Low[u] theo Number[v]}
end;
end;

procedure Solve;
var
  u, v: Integer;
begin
  Count := 0;
  FillChar(Parent, SizeOf(Parent), 0); {Đánh dấu mọi đỉnh đều chưa thăm}
  for u := 1 to n do
    if Parent[u] = 0 then {Gặp một đỉnh chưa thăm}
      begin
        Parent[u] := -1; {Cho u là một gốc cây DFS}
        Visit(u); {Xây dựng cây DFS gốc u}
      end;
  end;

procedure Result; {In kết quả}
var
  f: Text;
  u, v: Integer;
  nChildren: array[1..max] of Integer;
  IsCut: array[1..max] of Boolean;
begin
  FillChar(nChildren, SizeOf(nChildren), 0); {Tính nChildren[u] = Số nhánh con của nhánh DFS gốc u}
  for v := 1 to n do
    if Parent[v] <> -1 then Inc(nChildren[Parent[v]]);
  Assign(f, OutputFile); Rewrite(f);
  WriteLn(f, 'Bridges: ');
  for v := 1 to n do
    begin
      u := Parent[v];
      if (u <> -1) and (Low[v] >= Number[v]) then {(u, v) là cầu}
        WriteLn(f, '(', u, ', ', v, ')');
    end;
  WriteLn(f, 'Cut vertices:');
  FillChar(IsCut, SizeOf(IsCut), False);
  for v := 1 to n do
    if Parent[v] <> -1 then
      begin
        u := Parent[v];
        {Nếu Low[v] ≥ Number[u] khóp ⇔ u không phải gốc cây DFS hoặc u có ≥ 2 nhánh con}
        if (Low[v] >= Number[u]) then
          IsCut[u] := IsCut[u] or (Parent[u] <> -1) or (nChildren[u] >= 2);
      end;
  for u := 1 to n do
    if IsCut[u] then WriteLn(f, u);
  Close(f);
end;

begin
  Enter;
  Solve;
  Result;
end.


```


§6. CHU TRÌNH EULER, ĐƯỜNG ĐI EULER, ĐỒ THỊ EULER

6.1. BÀI TOÁN 7 CÁI CẦU

Thành phố Konigsberg thuộc Phổ (nay là Kaliningrad thuộc Cộng hòa Nga), được chia làm 4 vùng bằng các nhánh sông Pregel. Các vùng này gồm 2 vùng bên bờ sông (B, C), đảo Kneiphof (A) và một miền nằm giữa hai nhánh sông Pregel (D). Vào thế kỷ XVIII, người ta đã xây 7 chiếc cầu nối những vùng này với nhau. Người dân ở đây tự hỏi: Liệu có cách nào xuất phát tại một địa điểm trong thành phố, đi qua 7 chiếc cầu, mỗi chiếc đúng 1 lần rồi quay trở về nơi xuất phát không ?

Nhà toán học Thụy sĩ Leonhard Euler đã giải bài toán này và có thể coi đây là ứng dụng đầu tiên của Lý thuyết đồ thị, ông đã mô hình hóa sơ đồ 7 cái cầu bằng một đa đồ thị, bốn vùng được biểu diễn bằng 4 đỉnh, các cầu là các cạnh. Bài toán tìm đường qua 7 cầu, mỗi cầu đúng một lần có thể tổng quát hóa bằng bài toán: **Có tồn tại chu trình đơn trong đa đồ thị chưa tắt cả các cạnh ?.**



Hình 71: Mô hình đồ thị của bài toán bảy cái cầu

6.2. ĐỊNH NGHĨA

- ❖ Chu trình đơn chứa tất cả các cạnh của đồ thị được gọi là chu trình Euler
- ❖ Đường đi đơn chứa tất cả các cạnh của đồ thị được gọi là đường đi Euler
- ❖ Một đồ thị có chu trình Euler được gọi là đồ thị Euler
- ❖ Một đồ thị có đường đi Euler được gọi là đồ thị nửa Euler.

6.3. ĐỊNH LÝ

- ❖ Một đồ thị vô hướng liên thông $G = (V, E)$ có chu trình Euler khi và chỉ khi mọi đỉnh của nó đều có bậc chẵn: $\deg(v) \equiv 0 \pmod{2}$ ($\forall v \in V$)
- ❖ Một đồ thị vô hướng liên thông có đường đi Euler nhưng không có chu trình Euler khi và chỉ khi nó có đúng 2 đỉnh bậc lẻ
- ❖ Một đồ thi có hướng liên thông yếu $G = (V, E)$ có chu trình Euler thì mọi đỉnh của nó có bán bậc ra bằng bán bậc vào: $\deg^+(v) = \deg^-(v)$ ($\forall v \in V$); Ngược lại, nếu G liên thông yếu và mọi đỉnh của nó có bán bậc ra bằng bán bậc vào thì G có chu trình Euler, hay G sẽ là liên thông mạnh.

- ❖ Một đồ thị có hướng liên thông yếu $G = (V, E)$ có đường đi Euler nhưng không có chu trình Euler nếu tồn tại đúng hai đỉnh $u, v \in V$ sao cho $\deg^+(u) - \deg^-(u) = \deg^-(v) - \deg^+(v) = 1$, còn tất cả những đỉnh khác u và v đều có bán bậc ra bằng bán bậc vào.

6.4. THUẬT TOÁN FLEURY TÌM CHU TRÌNH EULER


6.4.1. Đối với đồ thị vô hướng liên thông, mọi đỉnh đều có bậc chẵn.

Xuất phát từ một đỉnh, ta chọn một cạnh liên thuộc với nó để đi tiếp theo hai nguyên tắc sau:

- ❖ Xoá bỏ cạnh đã đi qua
- ❖ Chỉ đi qua cầu khi không còn cạnh nào khác để chọn


Và ta cứ chọn cạnh đi một cách thoải mái như vậy cho tới khi không đi tiếp được nữa, đường đi tìm được là chu trình Euler.

Ví dụ: Với đồ thị ở Hình 72:



Hình 72

Nếu xuất phát từ đỉnh 1, có hai cách đi tiếp: hoặc sang 2 hoặc sang 3, giả sử ta sẽ sang 2 và xoá cạnh $(1, 2)$ vừa đi qua. Từ 2 chỉ có cách duy nhất là sang 4, nên cho dù $(2, 4)$ là cầu ta cũng phải đi sau đó xoá luôn cạnh $(2, 4)$. Đến đây, các cạnh còn lại của đồ thị có thể vẽ như Hình 73 bằng nét liền, các cạnh đã bị xoá được vẽ bằng nét đứt.



Hình 73

Bây giờ đang đứng ở đỉnh 4 thì ta có 3 cách đi tiếp: sang 3, sang 5 hoặc sang 6. Vì $(4, 3)$ là cầu nên ta sẽ không đi theo cạnh $(4, 3)$ mà sẽ đi $(4, 5)$ hoặc $(4, 6)$. Nếu đi theo $(4, 5)$ và cứ tiếp tục đi như vậy, ta sẽ được chu trình Euler là $\langle 1, 2, 4, 5, 7, 8, 6, 4, 3, 1 \rangle$. Còn đi theo $(4, 6)$ sẽ tìm được chu trình Euler là: $\langle 1, 2, 4, 6, 8, 7, 5, 4, 3, 1 \rangle$.

6.4.2. Đối với đồ thị có hướng liên thông yếu, mọi đỉnh đều có bán bậc ra bằng bán bậc vào.

Bằng cách “lạm dụng thuật ngữ”, ta có thể mô tả được thuật toán tìm chu trình Euler cho cả đồ thị có hướng cũng như vô hướng:

- ❖ Thứ nhất, dưới đây nếu ta nói cạnh (u, v) thì hiểu là cạnh nối đỉnh u và đỉnh v trên đồ thị vô hướng, hiểu là cung nối từ đỉnh u tới đỉnh v trên đồ thị có hướng.
- ❖ Thứ hai, ta gọi cạnh (u, v) là “một đi không trở lại” nếu như từ u ta đi tới v theo cạnh đó, sau đó xoá cạnh đó đi thì không có cách nào từ v quay lại u .

Vậy thì thuật toán Fleury tìm chu trình Euler có thể mô tả như sau:

Xuất phát từ một đỉnh, ta đi một cách tự ý theo các cạnh tuân theo hai nguyên tắc: Xoá bỏ cạnh vừa đi qua và chỉ chọn cạnh “một đi không trở lại” nếu như không còn cạnh nào khác để chọn.


6.5. CÀI ĐẶT

Ta sẽ cài đặt thuật toán Fleury trên một đa đồ thị vô hướng. Để đơn giản, ta coi đồ thị này đã có chu trình Euler, công việc của ta là tìm ra chu trình đó thôi. Bởi việc kiểm tra tính liên thông cũng như kiểm tra mọi đỉnh đều có bậc chẵn đến giờ có thể coi là chuyện nhỏ.

Input: file văn bản EULER.INP

- ❖ Dòng 1: Chứa số đỉnh n của đồ thị ($n \leq 100$)
- ❖ Các dòng tiếp theo, mỗi dòng chứa 3 số nguyên dương cách nhau ít nhất 1 dấu cách có dạng: $u \ v \ k$ cho biết giữa đỉnh u và đỉnh v có k cạnh nối

Output: file văn bản EULER.OUT, ghi chu trình Euler



P_4_06_1.PAS * Thuật toán Fleury tìm chu trình Euler

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Euler_Circuit;
const
  InputFile = 'EULER.INP';
  OutputFile = 'EULER.OUT';
  max = 100;
var
  a: array[1..max, 1..max] of Integer;
  n: Integer;

procedure Enter;
var
  u, v, k: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  
```

```

FillChar(a, SizeOf(a), 0);
ReadLn(f, n);
while not SeekEof(f) do
begin
  ReadLn(f, u, v, k);
  a[u, v] := k;
  a[v, u] := k;
end;
Close(f);
end;

{Thủ tục này kiểm tra nếu xoá một cạnh nối (x, y) thì y có còn quay lại được x hay không}
function CanGoBack(x, y: Integer): Boolean;
var
  Queue: array[1..max] of Integer; {Hàng đợi dùng cho Breadth First Search}
  Front, Rear: Integer; {Front: Chỉ số đầu hàng đợi, Rear: Chỉ số cuối hàng đợi}
  u, v: Integer;
  Free: array[1..max] of Boolean; {Mảng đánh dấu}
begin
  Dec(a[x, y]); Dec(a[y, x]); {Thủ xoá một cạnh (x, y)  $\Leftrightarrow$  Số cạnh nối (x, y) giảm 1}
  FillChar(Free, n, True); {sau đó áp dụng BFS để xem từ y có quay lại x được không ?}
  Free[y] := False;
  Front := 1; Rear := 1;
  Queue[1] := y;
repeat
  u := Queue[Front]; Inc(Front);
  for v := 1 to n do
    if Free[v] and (a[u, v] > 0) then
    begin
      Inc(Rear);
      Queue[Rear] := v;
      Free[v] := False;
      if Free[x] then Break;
    end;
  until Front > Rear;
  CanGoBack := not Free[x];
  Inc(a[x, y]); Inc(a[y, x]); {ở trên đã thủ xoá cạnh thì giờ phải phục hồi}
end;

procedure FindEulerCircuit; {Thuật toán Fleury}
var
  Current, Next, v, count: Integer;
  f: Text;
begin
  Assign(f, OutputFile); Rewrite(f);
  Current := 1;
  Write(f, 1, ' '); {Bắt đầu từ đỉnh Current = 1}
  count := 1;
repeat
  Next := 0;
  for v := 1 to n do
    if a[Current, v] > 0 then
    begin
      Next := v;
      if CanGoBack(Current, Next) then Break;
    end;
  if Next <> 0 then
  begin
    Dec(a[Current, Next]);
    Dec(a[Next, Current]); {Xoá bỏ cạnh vừa đi qua}
    Write(f, Next, ' '); {In kết quả đi tới Next}
    Inc(count);
    if count mod 16 = 0 then WriteLn; {In ra tối đa 16 đỉnh trên một dòng}
  end;
end;

```

```

Current := Next; {Lại tiếp tục với đỉnh đang đứng là Next}
end;
until Next = 0; {Cho tới khi không đi tiếp được nữa}
Close(f);
end;

begin
Enter;
FindEulerCircuit;
end.

```

6.6. THUẬT TOÁN TỐT HƠN

Trong trường hợp đồ thị Euler có **số cạnh đú nhỏ**, ta có thể sử dụng phương pháp sau để tìm chu trình Euler trong đồ thị vô hướng: Bắt đầu từ một chu trình đơn C bất kỳ, chu trình này tìm được bằng cách xuất phát từ một đỉnh, đi tuỳ ý theo các cạnh cho tới khi quay về đỉnh xuất phát, lưu ý là đi qua cạnh nào xoá luôn cạnh đó. Nếu như chu trình C tìm được chứa tất cả các cạnh của đồ thị thì đó là chu trình Euler. Nếu không, xét các đỉnh dọc theo chu trình C, nếu còn có cạnh chưa xoá liên thuộc với một đỉnh u nào đó thì lại từ u, ta đi tuỳ ý theo các cạnh cũng theo nguyên tắc trên cho tới khi quay trở về u, để được một chu trình đơn khác qua u. Loại bỏ vị trí u khỏi chu trình C và chèn vào C chu trình mới tìm được tại đúng vị trí của u vừa xoá, ta được một chu trình đơn C' mới lớn hơn chu trình C. Cứ làm như vậy cho tới khi được chu trình Euler. Việc chứng minh tính đúng đắn của thuật toán cũng là chứng minh định lý về điều kiện cần và đủ để một đồ thị vô hướng liên thông có chu trình Euler.

Mô hình thuật toán có thể viết như sau:

```

<Khởi tạo một ngăn xếp Stack ban đầu chỉ gồm mỗi đỉnh 1>;
<Viết các phương thức Push (đẩy vào) và Pop(lấy ra) một đỉnh từ ngăn xếp Stack, phương thức Get chỉ cho biết phần tử nằm ở đỉnh Stack. Khác với Pop, phương thức Get chỉ cho biết phần tử ở đỉnh Stack chứ không lấy phần tử đó ra>;
while Stack ≠ Ø do
begin
x := Get;
if ∃y: (x, y) ∈ E then {Từ x còn đi hướng khác được}
begin
Push(y);
<Loại bỏ cạnh (x, y) khỏi đồ thị>;
end
else {Từ x không đi tiếp được tới đâu nữa}
begin
x := Pop;
<In ra đỉnh x trên đường đi Euler>;
end;
end;

```

Thuật toán trên có thể dùng để tìm chu trình Euler trong đồ thị có hướng liên thông yếu, mọi đỉnh có bán bậc ra bằng bán bậc vào. Tuy nhiên thứ tự các đỉnh in ra bị ngược so với các cung định hướng, ta có thể đảo ngược hướng các cung trước khi thực hiện thuật toán để được thứ tự đúng.

Thuật toán hoạt động với hiệu quả cao, dễ cài đặt, nhưng trường hợp xấu nhất thì Stack sẽ phải chứa toàn bộ danh sách đỉnh trên chu trình Euler chính vì vậy mà khi đa đồ thị có số cạnh quá lớn thì có thể không đủ không gian nhớ mô tả Stack. Lý do thuật toán chỉ có thể áp dụng trong trường hợp số cạnh có giới hạn biết trước đủ nhỏ là như vậy.

P_4_06_2.PAS * Thuật toán hiệu quả tìm chu trình Euler

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Euler_Circuit;
const
  InputFile = 'EULER.INP';
  OutputFile = 'EULER.OUT';
  max = 100;
  maxE = 20000; {Số cạnh tối đa}
var
  a: array[1..max, 1..max] of Integer;
  stack: array[1..maxE] of Integer;
  n, Top: Integer;

procedure Enter; {Nhập dữ liệu}
var
  u, v, k: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  FillChar(a, SizeOf(a), 0);
  ReadLn(f, n);
  while not SeekEof(f) do
    begin
      ReadLn(f, u, v, k);
      a[u, v] := k;
      a[v, u] := k;
    end;
  Close(f);
end;

procedure Push(v: Integer); {Đẩy một đỉnh v vào ngăn xếp}
begin
  Inc(Top);
  Stack[Top] := v;
end;

function Pop: Integer; {Lấy một đỉnh khỏi ngăn xếp, trả về trong kết quả hàm}
begin
  Pop := Stack[Top];
  Dec(Top);
end;

function Get: Integer; {Trả về phần tử ở đỉnh (Top) ngăn xếp}
begin
  Get := Stack[Top];
end;

procedure FindEulerCircuit;
var
  u, v, count: Integer;
  f: Text;
begin
  Assign(f, OutputFile); Rewrite(f);
  Stack[1] := 1; {Khởi tạo ngăn xếp ban đầu chỉ gồm đỉnh 1}
  Top := 1;
  count := 0;
  while Top <> 0 do {Chừng nào ngăn xếp chưa rỗng}
    begin
      u := Get; {Xác định u là phần tử ở đỉnh ngăn xếp}
      for v := 1 to n do
        if a[u, v] > 0 then {Xét tất cả các cạnh liên thuộc với u, nếu thấy}
          begin
            Dec(a[u, v]); Dec(a[v, u]); {Xoá cạnh đó khỏi đồ thị}
            if a[u, v] > 0 then {Nếu còn cạnh, đẩy v vào ngăn xếp}
              begin
                Push(v);
                count := count + 1;
              end;
            end;
    end;
  end;
end;

```

```

Push(v); {Đẩy đỉnh tiếp theo vào ngăn xếp}
Break;
end;
if u = Get then {Nếu phần tử ở đỉnh ngăn xếp vẫn là u  $\Rightarrow$  vòng lặp trên không tìm thấy đỉnh nào kề với u}
begin
  Inc(count);
  Write(f, Pop, ' '); {In ra phần tử đỉnh ngăn xếp}
end;
end;
Close(f);
end;

begin
  Enter;
  FindEulerCircuit;
end.


```

Bài tập

Trên mặt phẳng cho n hình chữ nhật có các cạnh song song với các trục tọa độ. Hãy chỉ ra một chu trình:

Chỉ đi trên cạnh của các hình chữ nhật

Trên cạnh của mỗi hình chữ nhật, ngoại trừ những giao điểm với cạnh của hình chữ nhật khác có thể qua nhiều lần, những điểm còn lại chỉ được qua đúng một lần.



M D A B C M F G N L I J K N H E M

§7. CHU TRÌNH HAMILTON, ĐƯỜNG ĐI HAMILTON, ĐỒ THỊ HAMILTON


7.1. ĐỊNH NGHĨA

Cho đồ thị $G = (V, E)$ có n đỉnh

- ❖ Chu trình ($x[1], x[2], \dots, x[n], x[1]$) được gọi là chu trình Hamilton nếu $x[i] \neq x[j]$ với $\forall i, j: 1 \leq i < j \leq n$
- ❖ Đường đi ($x[1], x[2], \dots, x[n]$) được gọi là đường đi Hamilton nếu $x[i] \neq x[j]$ với $\forall i, j: 1 \leq i < j \leq n$
- ❖ Đồ thị có chu trình Hamilton được gọi là đồ thị Hamilton
- ❖ Đồ thị có đường đi Hamilton được gọi là đồ thị nửa Hamilton

Có thể phát biểu một cách hình thức: Chu trình Hamilton là chu trình xuất phát từ 1 đỉnh, đi thăm tất cả những đỉnh còn lại mỗi đỉnh đúng 1 lần, cuối cùng quay trở lại đỉnh xuất phát. Đường đi Hamilton là đường đi qua tất cả các đỉnh của đồ thị, mỗi đỉnh đúng 1 lần. Khác với khái niệm chu trình Euler và đường đi Euler, một chu trình Hamilton không phải là đường đi Hamilton bởi có đỉnh xuất phát được thăm tới 2 lần.

Ví dụ: Xét 3 đơn đồ thị G_1, G_2, G_3 như trong Hình 74:



Hình 74

Đồ thị G_1 có chu trình Hamilton (a, b, c, d, e, a). G_2 không có chu trình Hamilton vì $\deg(a) = 1$ nhưng có đường đi Hamilton (a, b, c, d). G_3 không có cả chu trình Hamilton lẫn đường đi Hamilton

7.2. ĐỊNH LÝ

- ❖ Đồ thị vô hướng G , trong đó tồn tại k đỉnh sao cho nếu xoá đi k đỉnh này cùng với những cạnh liên thuộc của chúng thì đồ thị nhận được sẽ có nhiều hơn k thành phần liên thông. Thì khẳng định là G không có chu trình Hamilton. Mệnh đề phản đảo của định lý này cho ta điều kiện cần để một đồ thị có chu trình Hamilton
- ❖ Định lý Dirac (1952): Đồ thị vô hướng G có n đỉnh ($n \geq 3$). Khi đó nếu mọi đỉnh v của G đều có $\deg(v) \geq n/2$ thì G có chu trình Hamilton. Đây là một điều kiện đủ để một đồ thị có chu trình Hamilton.

- ❖ Đồ thị có hướng G liên thông mạnh và có n đỉnh. Nếu $\deg^+(v) \geq n / 2$ và $\deg^-(v) \geq n / 2$ với mọi đỉnh v thì G có chu trình Hamilton


7.3. CÀI ĐẶT

Dưới đây ta sẽ cài đặt một chương trình liệt kê tất cả các chu trình Hamilton xuất phát từ đỉnh 1, các chu trình Hamilton khác có thể có được bằng cách hoán vị vòng quanh. Lưu ý rằng cho tới nay, người ta vẫn chưa tìm ra một phương pháp nào thực sự hiệu quả hơn phương pháp quay lui để tìm dù chỉ một chu trình Hamilton cũng như đường đi Hamilton trong trường hợp đồ thị tổng quát.

Input: file văn bản HAMILTON.INP

- ❖ Dòng 1 ghi số đỉnh n ($2 \leq n \leq 100$) và số cạnh m của đồ thị cách nhau 1 dấu cách
- ❖ m dòng tiếp theo, mỗi dòng có dạng hai số nguyên dương u, v cách nhau 1 dấu cách, thể hiện u, v là hai đỉnh kề nhau trong đồ thị

Output: file văn bản HAMILTON.OUT liệt kê các chu trình Hamilton



P_4_07_1.PAS * Thuật toán quay lui liệt kê chu trình Hamilton

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program All_of_Hamilton_Circuits;
const
  InputFile = 'HAMILTON.INP';
  OutputFile = 'HAMILTON.OUT';
  max = 100;
var
  fo: Text;
  a: array[1..max, 1..max] of Boolean; {Ma trận kề của đồ thị: a[u, v] = True ⇔ (u, v) là cạnh}
  Free: array[1..max] of Boolean; {Mảng đánh dấu Free[v] = True nếu chưa đi qua đỉnh v}
  x: array[1..max] of Integer; {Chu trình Hamilton sẽ tìm là: 1=x[1]→x[2] → ... →x[n] →x[1]=1}
  n: Integer;

procedure Enter;
var
  i, u, v, m: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  FillChar(a, SizeOf(a), False);
  ReadLn(f, n, m);
  for i := 1 to m do
    begin
      ReadLn(f, u, v);
      a[u, v] := True;
      a[v, u] := True;
    end;
  Close(f);
end;

```

```

procedure PrintResult; {In kết quả nếu tìm thấy chu trình Hamilton}
var
  i: Integer;
begin
  for i := 1 to n do Write(fo, x[i], ' ');
  WriteLn(fo, x[1]);
end;

procedure Attempt(i: Integer); {Thử các cách chọn đỉnh thứ i trong hành trình}
var
  j: Integer;
begin
  for j := 1 to n do {Đỉnh thứ i (x[i]) có thể chọn trong những đỉnh}
    if Free[j] and a[x[i - 1], j] then {kề với x[i - 1] và chưa bị đi qua}
      begin
        x[i] := j; {Thử một cách chọn x[i]}
        if i < n then {Nếu chưa thử chọn đến x[n]}
          begin
            Free[j] := False; {Đánh dấu đỉnh j là đã đi qua}
            Attempt(i + 1); {Để các bước thử kế tiếp không chọn phải đỉnh j nữa}
            Free[j] := True; {Sẽ thử phương án khác cho x[i] nên sẽ bỏ đánh dấu đỉnh vừa thử}
          end
        else {Nếu đã thử chọn đến x[n]}
          if a[j, x[1]] then PrintResult; {và nếu x[n] lại kề với x[1] thì ta có chu trình Hamilton}
        end;
      end;
begin
  Enter;
  FillChar(Free, SizeOf(Free), True); {Khởi tạo: Các đỉnh đều chưa đi qua}
  x[1] := 1; Free[1] := False; {Bắt đầu từ đỉnh 1}
  Assign(fo, OutputFile); Rewrite(fo);
  Attempt(2); {Thử các cách chọn đỉnh kế tiếp}
  Close(fo);
end.

```

Bài tập

Bài 1


- a) Lập chương trình nhập vào một đồ thị và chỉ ra đúng một chu trình Hamilton nếu có.
- b) Lập chương trình nhập vào một đồ thị và chỉ ra đúng một đường đi Hamilton nếu có.

Bài 2

Trong đám cưới của Persée và Andromède có $2n$ hiệp sỹ. Mỗi hiệp sỹ có không quá $n - 1$ kẻ thù. Hãy giúp Cassiopé, mẹ của Andromède xếp $2n$ hiệp sỹ ngồi quanh một bàn tròn sao cho không có hiệp sỹ nào phải ngồi cạnh kẻ thù của mình. Mỗi hiệp sỹ sẽ cho biết những kẻ thù của mình khi họ đến sân rồng.


Bài 3

Gray code: Một hình tròn được chia thành 2^n hình quạt đồng tâm. Hãy xếp tất cả các xâu nhị phân độ dài n vào các hình quạt, mỗi xâu vào một hình quạt sao cho bất cứ hai xâu nào ở hai hình quạt cạnh nhau đều chỉ khác nhau đúng 1 bit. Ví dụ với $n = 3$:



Bài 4

Thách đố: Bài toán mã đi tuần: Trên bàn cờ tổng quát kích thước $n \times n$ ô vuông (n chẵn và $6 \leq n \leq 20$). Trên một ô nào đó có đặt một quân mã. Quân mã đang ở ô $(x[1], y[1])$ có thể di chuyển sang ô $(x[2], y[2])$ nếu $|x[1]-x[2]| \cdot |y[1]-y[2]| = 2$ (Xem hình vẽ).



Hãy tìm một hành trình của quân mã từ ô xuất phát, đi qua tất cả các ô của bàn cờ, mỗi ô đúng 1 lần.

Ví dụ:

Với $n = 8$, ô xuất phát $(3, 3)$

45	42	3	18	35	20	5	8
2	17	44	41	4	7	34	21
43	46	1	36	19	50	9	6
16	31	48	59	40	33	22	51
47	60	37	32	49	58	39	10
30	15	64	57	38	25	52	23
61	56	13	28	63	54	11	26
14	29	62	55	12	27	24	53

Với $n = 10$, ô xuất phát $(6, 5)$

18	71	100	43	20	69	86	45	22	55
97	42	19	70	99	44	21	24	87	46
72	17	98	95	68	85	88	63	26	23
41	96	73	84	81	94	67	90	47	50
16	83	80	93	74	89	64	49	62	27
79	40	35	82	1	76	91	66	51	48
36	15	78	75	92	65	2	61	28	53
39	12	37	34	77	60	57	52	3	6
14	33	10	59	56	31	8	5	54	29
11	38	13	32	9	58	55	30	7	4

Hướng dẫn: Nếu coi các ô của bàn cờ là các đỉnh của đồ thị và các cạnh là nối giữa hai đỉnh tương ứng với hai ô mã giao chân thì dễ thấy rằng hành trình của quân mã cần tìm sẽ là một đường đi Hamilton. Ta có thể xây dựng hành trình bằng thuật toán quay lui kết hợp với phương pháp duyệt ưu tiên Warnsdorff: Nếu gọi $\text{deg}(x, y)$ là số ô kè với ô (x, y) và chưa đi qua (kè ở đây theo nghĩa đỉnh kè chứ không phải là ô kè cạnh) thì từ một ô ta sẽ không thử xét lần lượt các hướng đi có thể, mà ta sẽ ưu tiên thử hướng đi tới ô có deg nhỏ nhất trước. Trong trường hợp có tồn tại đường đi, phương pháp này hoạt động với tốc độ tuyệt vời: Với mọi n chẵn trong khoảng từ 6 tới 18, với mọi vị trí ô xuất phát, trung bình thời gian tính từ lúc bắt đầu tới lúc tìm ra một nghiệm < 1 giây. Tuy nhiên trong trường hợp n lẻ, có lúc không tồn tại đường đi, do phải duyệt hết mọi khả năng nên thời gian thực thi lại hết sức tồi tệ. (Có xét ưu tiên như trên hay xét thứ tự như trước kia thì cũng vậy thôi). Ta có thể thử với n lẻ: 5, 7, 9 ... và ô xuất phát (1, 2), sau đó ngồi xem máy tính toát mồ hôi).

§8. BÀI TOÁN ĐƯỜNG ĐI NGẮN NHẤT

8.1. ĐỒ THỊ CÓ TRỌNG SỐ

Đồ thị mà mỗi cạnh của nó được gán cho tương ứng với một số (nguyên hoặc thực) được gọi là đồ thị có trọng số. Số gán cho mỗi cạnh của đồ thị được gọi là trọng số của cạnh. Tương tự như đồ thị không trọng số, có nhiều cách biểu diễn đồ thị có trọng số trong máy tính. Đối với đơn đồ thị thì cách dễ dàng nhất là sử dụng ma trận trọng số:

Giả sử đồ thị $G = (V, E)$ có n đỉnh. Ta sẽ dựng ma trận vuông C kích thước $n \times n$. Ở đây:

- ❖ Nếu $(u, v) \in E$ thì $C[u, v] =$ trọng số của cạnh (u, v)
- ❖ Nếu $(u, v) \notin E$ thì tùy theo trường hợp cụ thể, $C[u, v]$ được gán một giá trị nào đó để có thể nhận biết được (u, v) không phải là cạnh (Chẳng hạn có thể gán bằng $+\infty$, hay bằng 0, bằng $-\infty$, v.v...)
- ❖ Quy ước $c[v, v] = 0$ với mọi đỉnh v .

Đường đi, chu trình trong đồ thị có trọng số cũng được định nghĩa giống như trong trường hợp không trọng số, chỉ có khác là độ dài đường đi không phải tính bằng số cạnh đi qua, mà được tính bằng tổng trọng số của các cạnh đi qua.

8.2. BÀI TOÁN ĐƯỜNG ĐI NGẮN NHẤT

Trong các ứng dụng thực tế, chẳng hạn trong mạng lưới giao thông đường bộ, đường thuỷ hoặc đường không. Người ta không chỉ quan tâm đến việc tìm đường đi giữa hai địa điểm mà còn phải lựa chọn một hành trình tiết kiệm nhất (theo tiêu chuẩn không gian, thời gian hay chi phí). Khi đó phát sinh yêu cầu tìm đường đi ngắn nhất giữa hai đỉnh của đồ thị. Bài toán đó phát biểu dưới dạng tổng quát như sau: Cho đồ thị có trọng số $G = (V, E)$, hãy tìm một đường đi ngắn nhất từ đỉnh xuất phát $s \in V$ đến đỉnh đích $f \in V$. Độ dài của đường đi này ta sẽ ký hiệu là $d[s, f]$ và gọi là **khoảng cách** (distance) từ s đến f . Nếu như không tồn tại đường đi từ s tới f thì ta sẽ đặt khoảng cách đó $= +\infty$.

Nếu như đồ thị có chu trình âm (chu trình với độ dài âm) thì khoảng cách giữa một số cặp đỉnh nào đó có thể không xác định, bởi vì bằng cách đi vòng theo chu trình này một số lần đủ lớn, ta có thể chỉ ra đường đi giữa hai đỉnh nào đó trong chu trình này nhỏ hơn bất kỳ một số cho trước nào. Trong trường hợp như vậy, có thể đặt vấn đề tìm **đường đi cơ bản** (đường đi không có đỉnh lặp lại) ngắn nhất. Vấn đề đó là một vấn đề hết sức phức tạp mà ta sẽ không bàn tới ở đây.

Nếu như đồ thị không có chu trình âm thì ta có thể chứng minh được rằng một trong những đường đi ngắn nhất là đường đi cơ bản. Và nếu như biết được khoảng cách từ s tới tất cả những đỉnh khác thì đường đi ngắn nhất từ s tới f có thể tìm được một cách dễ dàng qua thuật toán sau:

Gọi $c[u, v]$ là trọng số của cạnh $[u, v]$. Qui ước $c[v, v] = 0$ với mọi $v \in V$ và $c[u, v] = +\infty$ nếu như $(u, v) \notin E$. Đặt $d[s, v]$ là khoảng cách từ s tới v . Để tìm đường đi từ s tới f , ta có thể nhận thấy rằng luôn tồn tại đỉnh $f_1 \neq f$ sao cho:


$$d[s, f] = d[s, f_1] + c[f_1, f]$$

(Độ dài đường đi ngắn nhất $s \rightarrow f$ = Độ dài đường đi ngắn nhất $s \rightarrow f_1$ + Chi phí đi từ $f_1 \rightarrow f$)

Đỉnh f_1 đó là đỉnh liền trước f trong đường đi ngắn nhất từ s tới f . Nếu $f_1 = s$ thì đường đi ngắn nhất là đường đi trực tiếp theo cung (s, f) . Nếu không thì vẫn đề trở thành tìm đường đi ngắn nhất từ s tới f_1 . Và ta lại tìm được một đỉnh f_2 khác f và f_1 để:

$$d[s, f_1] = d[s, f_2] + c[f_2, f_1]$$

Cứ tiếp tục như vậy, sau một số hữu hạn bước, ta suy ra rằng dãy f, f_1, f_2, \dots không chứa đỉnh lặp lại và kết thúc ở s . Lật ngược thứ tự dãy cho ta đường đi ngắn nhất từ s tới f .




Tuy nhiên, người ta thường không sử dụng phương pháp này mà sẽ kết hợp lưu vết đường đi ngay trong quá trình tìm kiếm.

Dưới đây ta sẽ xét một số thuật toán tìm đường đi ngắn nhất từ đỉnh s tới đỉnh f trên đồ thị có hướng $G = (V, E)$ có n đỉnh và m cung. Trong trường hợp đơn đồ thị vô hướng với trọng số không âm, bài toán tìm đường đi ngắn nhất có thể dẫn về bài toán trên đồ thị có hướng bằng cách thay mỗi cạnh của nó bằng hai cung có hướng ngược chiều nhau. Bạn có thể đưa vào một số sửa đổi nhỏ trong thủ tục nhập liệu để giải quyết bài toán trong trường hợp đa đồ thị.

Input: file văn bản MINPATH.INP

- ❖ Dòng 1: Chứa số đỉnh n (≤ 1000), số cung m của đồ thị, đỉnh xuất phát s , đỉnh đích f cách nhau ít nhất 1 dấu cách
- ❖ m dòng tiếp theo, mỗi dòng có dạng ba số $u, v, c[u, v]$ cách nhau ít nhất 1 dấu cách, thể hiện (u, v) là một cung $\in E$ và trọng số của cung đó là $c[u, v]$ ($c[u, v]$ là số nguyên có giá trị tuyệt đối ≤ 1000)

Output: file văn bản MINPATH.OUT ghi đường đi ngắn nhất từ s tới f và độ dài đường đi đó



8.3. TRƯỜNG HỢP ĐỒ THỊ KHÔNG CÓ CHU TRÌNH ÂM - THUẬT TOÁN FORD-BELLMAN

Thuật toán Ford-Bellman có thể phát biểu rất đơn giản:

Với đỉnh xuất phát s. Gọi $d[v]$ là khoảng cách từ s tới v với các giá trị khởi tạo là:

- ❖ $d[s] := 0$
- ❖ $d[v] := +\infty$ nếu $v \neq s$

Sau đó ta tối ưu hoá dần các $d[v]$ như sau: Xét mọi cặp đỉnh u, v của đồ thị, nếu có một cặp đỉnh u, v mà $d[v] > d[u] + c[u, v]$ thì ta đặt lại $d[v] := d[u] + c[u, v]$. Tức là nếu độ dài đường đi từ s tới v lại **lớn hơn** tổng độ dài đường đi từ s tới u cộng với chi phí đi từ u tới v thì ta sẽ huỷ bỏ đường đi từ s tới v đang có và coi đường đi từ s tới v chính là đường đi từ s tới u sau đó đi tiếp từ u tới v. Chú ý rằng ta đặt $c[u, v] = +\infty$ nếu (u, v) không là cung. Thuật toán sẽ kết thúc khi không thể tối ưu thêm bất kỳ một nhãn $d[v]$ nào nữa.

```

for ( $\forall v \in V$ ) do  $d[v] := +\infty$ ;
 $d[s] := 0$ ;
repeat
  Stop := True;
  for ( $\forall u \in V$ ) do
    for ( $\forall v \in V: (u, v) \in E$ ) do
      if  $d[v] > d[u] + c[u, v]$  then
        begin
           $d[v] := d[u] + c[u, v]$ ;
          Stop := False;
        end;
    until Stop;
  
```

Tính đúng của thuật toán:

Tại bước khởi tạo thì mỗi $d[v]$ chính là độ dài ngắn nhất của đường đi từ s tới v qua không quá 0 cạnh.

Giả sử khi bắt đầu bước lặp thứ i ($i \geq 1$), $d[v]$ đã bằng độ dài đường đi ngắn nhất từ s tới v qua không quá $i - 1$ cạnh. Bởi đường đi từ s tới v qua không quá $i - 1$ cạnh sẽ phải thành lập bằng cách: lấy một đường đi từ s tới một đỉnh u nào đó qua không quá $i - 1$ cạnh, rồi đi tiếp tới v bằng cung (u, v) , nên độ dài đường đi ngắn nhất từ s tới v qua không quá $i - 1$ cạnh sẽ được tính bằng giá trị nhỏ nhất trong các giá trị (Nguyên lý tối ưu Bellman):

- ❖ Độ dài đường đi ngắn nhất từ s tới v qua không quá $i - 1$ cạnh

- ❖ Độ dài đường đi ngắn nhất từ s tới u qua không quá i - 1 cạnh cộng với trọng số cạnh (u, v)
 $(\forall u)$

Vì vậy, sau bước lặp tối ưu các $d[v]$ bằng công thức

$$d[v]_{\text{bước } i} = \min(d[v]_{\text{bước } i-1}, d[u]_{\text{bước } i-1} + c[u, v]) (\forall u)$$

thì các $d[v]$ sẽ bằng độ dài đường đi ngắn nhất từ s tới v qua không quá i cạnh.

Sau bước lặp tối ưu thứ n - 1, ta có $d[v] =$ độ dài đường đi ngắn nhất từ s tới v qua không quá n - 1 cạnh. Vì đồ thị không có chu trình âm nên sẽ có một đường đi ngắn nhất từ s tới v là đường đi cơ bản (qua không quá n - 1 cạnh). Tức là $d[v]$ sẽ là độ dài đường đi ngắn nhất từ s tới v.

Vậy thì số bước lặp tối ưu hoá sẽ không quá n - 1 bước.

Trong khi cài đặt chương trình, nếu mỗi bước lặp được mô tả dưới dạng:

```
for u := 1 to n do
  for v := 1 to n do
    d[v] := min(d[v], d[u] + c[u, v]);
```

Sự tối ưu bắc cầu (dùng $d[u]$ tối ưu $d[v]$ rồi lại có thể dùng $d[v]$ tối ưu $d[w]$ nữa...) chỉ làm tốc độ tối ưu các nhãn $d[.]$ tăng nhanh hơn nên số bước lặp vẫn sẽ không quá n - 1 bước

P_4_08_1.PAS * Thuật toán Ford-Bellman

```
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
```

```
program Shortest_Path_by_Ford_Bellman;
const
```

```
  InputFile = 'MINPATH.INP';
```

```
  OutputFile = 'MINPATH.OUT';
```

```
  max = 1000;
```

```
  maxEC = 1000;
```

```
  maxC = max * maxEC;
```

```
var
```

```
  c: array[1..max, 1..max] of Integer;
```

```
  d: array[1..max] of Integer;
```

```
  Trace: array[1..max] of Integer;
```

```
  n, s, f: Integer;
```

```
procedure LoadGraph; {Nhập đồ thị, đồ thị không được có chu trình âm}
```

```
var
```

```
  i, m, u, v: Integer;
```

```
  fi: Text;
```

```
begin
```

```
  Assign(fi, InputFile); Reset(fi);
```

```
  ReadLn(fi, n, m, s, f);
```

```
  {Những cạnh không có trong đồ thị được gán trọng số +∞}
```

```
  for u := 1 to n do
```

```
    for v := 1 to n do
```

```
      if u = v then c[u, v] := 0 else c[u, v] := maxC;
```

```
  for i := 1 to m do ReadLn(fi, u, v, c[u, v]);
```

```
  Close(fi);
```

```
end;
```

```
procedure Init; {Khởi tạo}
```

```
var
```

```
  i: Integer;
```

```
begin
```

```
  for i := 1 to n do d[i] := MaxC;
```

```
  d[s] := 0;
```

```

end;

procedure Ford_Bellman; {Thuật toán Ford-Bellman}
var
  Stop: Boolean;
  u, v, CountLoop: Integer;
begin
  for CountLoop := 1 to n - 1 do
    begin
      Stop := True;
      for u := 1 to n do
        for v := 1 to n do
          if d[v] > d[u] + c[u, v] then {Nếu  $\exists u, v$  thoả mãn  $d[v] > d[u] + c[u, v]$  thì tối ưu lại  $d[v]$ }
            begin
              d[v] := d[u] + c[u, v];
              Trace[v] := u; {Lưu vết đường đi}
              Stop := False;
            end;
      if Stop then Break;
    end;
  {Thuật toán kết thúc khi không sửa nhãn các  $d[v]$  được nữa hoặc đã lặp đủ  $n - 1$  lần}
end;

procedure PrintResult; {In đường đi từ s tới f}
var
  fo: Text;
begin
  Assign(fo, OutputFile); Rewrite(fo);
  if d[f] = maxC then {Nếu  $d[f]$  vẫn là  $+\infty$  thì tức là không có đường}
    writeln(fo, 'There is no path from ', s, ' to ', f)
  else {Truy vết tìm đường đi}
    begin
      writeln(fo, 'Distance from ', s, ' to ', f, ': ', d[f]);
      while f <> s do
        begin
          write(fo, f, '<-');
          f := Trace[f];
        end;
      writeln(fo, s);
    end;
  Close(fo);
end;

begin
  LoadGraph;
  Init;
  Ford_Bellman;
  PrintResult;
end.

```

8.4. TRƯỜNG HỢP TRỌNG SỐ TRÊN CÁC CUNG KHÔNG ÂM - THUẬT TOÁN DIJKSTRA

Trong trường hợp trọng số trên các cung không âm, thuật toán do Dijkstra đề xuất dưới đây hoạt động hiệu quả hơn nhiều so với thuật toán Ford-Bellman. Ta hãy xem trong trường hợp này, thuật toán Ford-Bellman thiếu hiệu quả ở chỗ nào:

Với đỉnh $v \in V$, Gọi $d[v]$ là độ dài đường đi ngắn nhất từ s tới v . Thuật toán Ford-Bellman khởi gán $d[s] = 0$ và $d[v] = +\infty$ với $\forall v \neq s$, sau đó tối ưu hoá dần các nhãn $d[v]$ bằng cách sửa

nhân theo công thức: $d[v] := \min(d[v], d[u] + c[u, v])$ với $\forall u, v \in V$. Như vậy nếu như ta dùng đỉnh u sửa nhân đỉnh v , sau đó nếu ta lại tối ưu được $d[u]$ thêm nữa thì ta cũng phải sửa lại nhân $d[v]$ dẫn tới việc $d[v]$ có thể phải chỉnh đi chỉnh lại rất nhiều lần. Vậy nên chăng, tại mỗi bước **không phải ta xét mọi cặp đỉnh (u, v)** để dùng đỉnh u sửa nhân đỉnh v mà sẽ **chọn đỉnh u là đỉnh mà không thể tối ưu nhân $d[u]$ thêm được nữa**.

Thuật toán Dijkstra (E.Dijkstra - 1959) có thể mô tả như sau:

Bước 1: Khởi tạo

Với đỉnh $v \in V$, gọi nhân $d[v]$ là độ dài đường đi ngắn nhất từ s tới v . Ban đầu $d[v]$ được khởi gán như trong thuật toán Ford-Bellman ($d[s] = 0$ và $d[v] = \infty$ với $\forall v \neq s$). Nhân của mỗi đỉnh có hai trạng thái tự do hay cố định, nhân tự do có nghĩa là có thể còn tối ưu hơn được nữa và nhân cố định tức là $d[v]$ đã bằng độ dài đường đi ngắn nhất từ s tới v nên không thể tối ưu thêm. Để làm điều này ta có thể sử dụng kỹ thuật đánh dấu: $Free[v] = \text{TRUE}$ hay FALSE tuỳ theo $d[v]$ tự do hay cố định. Ban đầu các nhân đều tự do.

Bước 2: Lặp

Bước lặp gồm có hai thao tác:

- ❖ Cố định nhân: Chọn trong các đỉnh có nhân tự do, lấy ra đỉnh u là đỉnh có $d[u]$ nhỏ nhất, và cố định nhân đỉnh u .
- ❖ Sửa nhân: Dùng đỉnh u , xét tất cả những đỉnh v và sửa lại các $d[v]$ theo công thức:

$$d[v] := \min(d[v], d[u] + c[u, v])$$

Bước lặp sẽ kết thúc khi mà đỉnh đích f được cố định nhân (tìm được đường đi ngắn nhất từ s tới f); hoặc tại thao tác cố định nhân, tất cả các đỉnh tự do đều có nhân là $+\infty$ (không tồn tại đường đi). Có thể đặt câu hỏi, ở thao tác 1, tại sao đỉnh u như vậy được cố định nhân, giả sử $d[u]$ còn có thể tối ưu thêm được nữa thì tất phải có một đỉnh t mang nhân tự do sao cho $d[u] > d[t] + c[t, u]$. Do trọng số $c[t, u]$ không âm nên $d[u] > d[t]$, trái với cách chọn $d[u]$ là nhỏ nhất. Tuy nhiên trong lần lặp đầu tiên thì s là đỉnh được cố định nhân do $d[s] = 0$.

Bước 3: Kết hợp với việc lưu vết đường đi trên từng bước sửa nhân, thông báo đường đi ngắn nhất tìm được hoặc cho biết không tồn tại đường đi ($d[f] = +\infty$).

```
for (v ∈ V) do d[v] := +∞;
d[s] := 0;
repeat
  u := arg min(d[v] | v ∈ V); {Lấy u là đỉnh có nhân d[u] nhỏ nhất}
  if (u = f) or (d[u] = +∞) then Break; {Hoặc tìm ra đường đi ngắn nhất từ s tới f, hoặc kết luận không có đường}
  for (v ∈ V: (u, v) ∈ E) do {Dùng u tối ưu nhân những đỉnh v kề với u}
    d[v] := min (d[v], d[u] + c[u, v]);
until False;
```

P_4_08_2.PAS * Thuật toán Dijkstra

```
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)  
program Shortest_Path_by_Dijkstra;  
const  
  InputFile = 'MINPATH.INP';
```

```

OutputFile = 'MINPATH.OUT';
max = 1000;
maxEC = 1000;
maxC = max * maxEC;
var
  c: array[1..max, 1..max] of Integer;
  d: array[1..max] of Integer;
  Trace: array[1..max] of Integer;
  Free: array[1..max] of Boolean; {Free[u] = True ⇔ u có nhãn tự do}
  n, s, f: Integer;

procedure LoadGraph; {Nhập đồ thị, trọng số các cung phải là số không âm}
var
  i, m, u, v: Integer;
  fi: Text;
begin
  Assign(fi, InputFile); Reset(fi);
  ReadLn(fi, n, m, s, f);
  for u := 1 to n do
    for v := 1 to n do
      if u = v then c[u, v] := 0 else c[u, v] := maxC;
  for i := 1 to m do ReadLn(fi, u, v, c[u, v]);
  Close(fi);
end;

procedure Init; {Khởi tạo các nhãn d[v], các đỉnh đều được coi là tự do}
var
  i: Integer;
begin
  for i := 1 to n do d[i] := MaxC;
  d[s] := 0;
  FillChar(Free, SizeOf(Free), True);
end;

procedure Dijkstra; {Thuật toán Dijkstra}
var
  i, u, v: Integer;
  min: Integer;
begin
  repeat
    {Tìm trong các đỉnh có nhãn tự do ra đỉnh u có d[u] nhỏ nhất}
    u := 0; min := maxC;
    for i := 1 to n do
      if Free[i] and (d[i] < min) then
        begin
          min := d[i];
          u := i;
        end;
  until False;
  {Thuật toán sẽ kết thúc khi các đỉnh tự do đều có nhãn +∞ hoặc đã chọn đến đỉnh f}
  if (u = 0) or (u = f) then Break;
  {Có định nhãn đỉnh u}
  Free[u] := False;
  {Dùng đỉnh u tối ưu nhãn những đỉnh tự do kề với u}
  for v := 1 to n do
    if Free[v] and (d[v] > d[u] + c[u, v]) then
      begin
        d[v] := d[u] + c[u, v];
        Trace[v] := u;
      end;
  until False;
end;

procedure PrintResult; {In đường đi từ s tới f}

```

```

var
  fo: Text;
begin
  Assign(fo, OutputFile); Rewrite(fo);
  if d[f] = maxC then
    WriteLn(fo, 'There is no path from ', s, ' to ', f)
  else
    begin
      WriteLn(fo, 'Distance from ', s, ' to ', f, ': ', d[f]);
      while f <> s do
        begin
          Write(fo, f, '<-');
          f := Trace[f];
        end;
      WriteLn(fo, s);
    end;
  Close(fo);
end;

begin
  LoadGraph;
  Init;
  Dijkstra;
  PrintResult;
end.

```

8.5. THUẬT TOÁN DIJKSTRA VÀ CẤU TRÚC HEAP

Nếu đồ thị thưa (có nhiều đỉnh, ít cạnh) ta có thể sử dụng danh sách kè kèm trọng số để biểu diễn đồ thị, tuy nhiên tốc độ của thuật toán Dijkstra vẫn khá chậm vì trong trường hợp xấu nhất, nó cần n lần cố định nhãn và mỗi lần tìm đỉnh để cố định nhãn sẽ mất một đoạn chương trình với độ phức tạp $O(n)$. Để thuật toán làm việc hiệu quả hơn, người ta thường sử dụng cấu trúc dữ liệu Heap (PHẦN 2, §8, 8.7.1) để lưu các đỉnh chưa cố định nhãn. Heap ở đây là một cây nhị phân hoàn chỉnh thoả mãn: Nếu u là đỉnh lưu ở nút cha và v là đỉnh lưu ở nút con thì $d[u] \leq d[v]$. (Đỉnh r lưu ở gốc Heap là đỉnh có $d[r]$ nhỏ nhất).

Tại mỗi bước lặp của thuật toán Dijkstra có hai thao tác: Tìm đỉnh cố định nhãn và Sửa nhãn.

- ❖ Với mỗi đỉnh v, gọi $\text{Pos}[v]$ là vị trí đỉnh v trong Heap, quy ước $\text{Pos}[v] = 0$ nếu v chưa bị đẩy vào Heap. Mỗi lần có thao tác sửa đổi vị trí các đỉnh trên cấu trúc Heap, ta lưu ý cập nhập lại mảng Pos này.
- ❖ Thao tác tìm đỉnh cố định nhãn sẽ lấy đỉnh lưu ở gốc Heap, cố định nhãn, đưa phần tử cuối Heap vào thế chỗ và thực hiện việc vun đồng (Adjust).
- ❖ Thao tác sửa nhãn, sẽ duyệt danh sách kè của đỉnh vừa cố định nhãn và sửa nhãn những đỉnh tự do kè với đỉnh này, mỗi lần sửa nhãn một đỉnh nào đó, ta xác định đỉnh này nằm ở đâu trong Heap (dựa vào mảng Pos) và thực hiện việc chuyển đỉnh đó lên (UpHeap) phía gốc Heap nếu cần để bảo toàn cấu trúc Heap.

Cài đặt dưới đây có Input/Output giống như trên nhưng có thể thực hiện trên đồ thị 10000 đỉnh, 100000 cạnh, trọng số mỗi cạnh ≤ 100000 .

```

P_4_08_3.PAS * Thuật toán Dijkstra và cấu trúc Heap

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Shortest_Path_by_Dijkstra_and_Heap;
const
  InputFile = 'MINPATH.INP';
  OutputFile = 'MINPATH.OUT';
  max = 10000;
  maxE = 100000;
  maxEC = 100000;
  maxC = max * maxEC;
type
  TAdj = array[1..maxE] of Integer;
  TAdjCost = array[1..maxE] of Integer;
  THeader = array[1..max + 1] of Integer;
var
  adj: TAdj; {Danh sách kề dạng mảng}
  adjCost: TAdjCost; {Kèm trọng số}
  h: THeader; {Mảng đánh dấu các đoạn trong danh sách kề adj}
  d: array[1..max] of Integer;
  Trace: array[1..max] of Integer;
  Free: array[1..max] of Boolean;
  heap: array[1..max] of Integer; {heap[i] = đỉnh lưu tại nút i của heap}
  Pos: array[1..max] of Integer; {pos[v] = vị trí của nút v trong heap (tức là pos[heap[i]] = i)}
  n, s, f, nHeap: Integer;

procedure LoadGraph; {Nhập dữ liệu}
var
  i, m, u, v, c: Integer;
  fi: Text;
begin
  {Đọc file lần 1, để xác định các đoạn}
  Assign(fi, InputFile); Reset(fi);
  ReadLn(fi, n, m, s, f);
  {Phép đếm phân phối (Distribution Counting)}
  FillChar(h, SizeOf(h), 0);
  for i := 1 to m do
    begin
      ReadLn(fi, u); {Ta chỉ cần tính bán bậc ra (deg+) của mỗi đỉnh nên không cần đọc đủ 3 thành phần}
      Inc(h[u]);
    end;
  for i := 2 to n do h[i] := h[i - 1] + h[i];
  Close(fi);
  {Đến đây, ta xác định được h[u] là vị trí cuối của danh sách kề đỉnh u trong adj}
  Reset(fi); {Đọc file lần 2, vào cấu trúc danh sách kề}
  ReadLn(fi); {Bỏ qua dòng đầu tiên Input file}
  for i := 1 to m do
    begin
      ReadLn(fi, u, v, c);
      adj[h[u]] := v; {Điền v và c vào vị trí đúng trong danh sách kề của u}
      adjCost[h[u]] := c;
      Dec(h[u]);
    end;
  h[n + 1] := m;
  Close(fi);
end;

procedure Init; {Khởi tạo d[i] = độ dài đường đi ngắn nhất từ s tới i qua 0 cạnh, Heap rỗng}
var
  i: Integer;
begin
  for i := 1 to n do d[i] := maxC;
  d[s] := 0;
  FillChar(Free, SizeOf(Free), True);

```

```

FillChar(Pos, SizeOf(Pos), 0);
nHeap := 0;
end;

procedure Update(v: Integer); {Định v vừa được sửa nhẫn, cần phải chỉnh lại Heap}
var
  parent, child: Integer;
begin
  child := Pos[v]; {child là vị trí của v trong Heap}
  if child = 0 then {Nếu v chưa có trong Heap thì Heap phải bổ sung thêm 1 phần tử và coi child = nút lá cuối Heap}
    begin
      Inc(nHeap); child := nHeap;
    end;
  parent := child div 2; {parent là nút cha của child}
  while (parent > 0) and (d[heap[parent]] > d[v]) do
    begin {Nếu đỉnh lưu ở nút parent ưu tiên kém hơn v thì đỉnh đó sẽ bị đẩy xuống nút con child}
      heap[child] := heap[parent]; {Đẩy đỉnh lưu trong nút cha xuống nút con}
      Pos[heap[child]] := child; {Ghi nhận lại vị trí mới của đỉnh đó}
      child := parent; {Tiếp tục xét lên phía nút gốc}
      parent := child div 2;
    end;
  {Thao tác "kéo xuống" ở trên tạo ra một "khoảng trống" tại nút child của Heap, đỉnh v sẽ được đặt vào đây}
  heap[child] := v;
  Pos[v] := child;
end;

function Pop: Integer; {Lấy từ Heap ra đỉnh có nhẫn tự do nhỏ nhất}
var
  r, c, v: Integer;
begin
  r := heap[1]; {Nút gốc Heap chứa đỉnh có nhẫn tự do nhỏ nhất}
  v := heap[nHeap]; {v là đỉnh ở nút lá cuối Heap, sẽ được đào lên đầu và vun đồng}
  Dec(nHeap);
  r := 1; {Bắt đầu từ nút gốc}
  while r * 2 <= nHeap do {Chừng nào r chưa phải là lá}
    begin
      {Chọn c là nút chứa đỉnh ưu tiên hơn trong hai nút con}
      c := r * 2;
      if (c < nHeap) and (d[heap[c + 1]] < d[heap[c]]) then Inc(c);
      {Nếu v ưu tiên hơn cả đỉnh chứa trong c, thì thoát ngay}
      if d[v] <= d[heap[c]] then Break;
      heap[r] := heap[c]; {Chuyển đỉnh lưu ở nút con c lên nút cha r}
      Pos[heap[r]] := r; {Ghi nhận lại vị trí mới trong Heap của đỉnh đó}
      r := c; {Gán nút cha := nút con và lặp lại}
    end;
  heap[r] := v; {Định v sẽ được đặt vào nút r để bảo toàn cấu trúc Heap}
  Pos[v] := r;
end;

procedure Dijkstra;
var
  i, u, iv, v, min: Integer;
begin
  Update(s); {Đưa đỉnh xuất phát về gốc Heap}
  repeat
    u := Pop; {Chọn đỉnh tự do có nhẫn nhỏ nhất}
    if u = f then Break; {Nếu đỉnh đó là f thì dừng ngay}
    Free[u] := False; {Cô định nhẫn đỉnh đó}
    for iv := h[u] + 1 to h[u + 1] do {Xét danh sách kề}
      begin
        v := adj[iv];
        if Free[v] and (d[v] > d[u] + adjCost[iv]) then
          begin

```

```

d[v] := d[u] + adjCost[iv]; {Tối ưu hóa nhãn của các đỉnh tự do kề với u}
Trace[v] := u; {Lưu vết đường đi}
Update(v); {Tổ chức lại Heap}
end;
end;
until nHeap = 0; {Không còn đỉnh nào mang nhãn tự do}
end;


procedure PrintResult;
var
  fo: Text;
begin
  Assign(fo, OutputFile); Rewrite(fo);
  if d[f] = maxC then
    WriteLn(fo, 'There is no path from ', s, ' to ', f)
  else
    begin
      WriteLn(fo, 'Distance from ', s, ' to ', f, ': ', d[f]);
      while f <> s do
        begin
          Write(fo, f, '<-');
          f := Trace[f];
        end;
      WriteLn(fo, s);
    end;
  Close(fo);
end;

begin
  LoadGraph;
  Init;
  Dijkstra;
  PrintResult;
end.

```

8.6. TRƯỜNG HỢP ĐỒ THỊ KHÔNG CÓ CHU TRÌNH - SẮP XẾP TÔ PÔ

Xét trường hợp đồ thị có hướng, không có chu trình (Directed Acyclic Graph - DAG), ta có một thuật toán hiệu quả dựa trên kỹ thuật sắp xếp Tô pô (Topological Sorting), cơ sở của thuật toán dựa vào định lý: Nếu $G = (V, E)$ là một DAG thì các đỉnh của nó có thể đánh số sao cho mỗi cung của G chỉ nối từ đỉnh có chỉ số nhỏ hơn đến đỉnh có chỉ số lớn hơn.



Hình 75: Phép đánh lại chỉ số theo thứ tự tốpô

Để đánh số lại các đỉnh theo điều kiện trên, ta có thể dùng thuật toán tìm kiếm theo chiều sâu và đánh số ngược lại với thứ tự duyệt xong, có thể viết như sau:

```

procedure Number;
procedure Visit(u: Integer);

```

```

var
  v: Integer;
begin
  {Đánh dấu u đã thăm};
  for (All v ∈ V: v chưa thăm kè với u) do Visit(v); {Thăm tiếp những đỉnh chưa thăm kè với u}
  {Đã duyệt xong nhánh DFS gốc u, đánh số mới cho đỉnh u là count};
  count := count - 1;
end;

begin
  {Đánh dấu mọi đỉnh ∈ V đều chưa thăm};
  count := n; {Biến đánh số được khởi tạo bằng n để đếm lùi}
  for u := 1 to n do
    if (u chưa thăm) then Visit(u);
end;

```

Việc kiểm tra đồ thị không được có chương trình cũng rất đơn giản bằng cách thêm vào đoạn chương trình trên vài dòng lệnh, tôi sẽ không viết dài để tập trung vào thuật toán, các bạn có thể tham khảo các kỹ thuật trình bày trong thuật toán Tarjan (§4, 4.4.3).

Nếu các đỉnh được đánh số sao cho mỗi cung phải nối từ một đỉnh tới một đỉnh khác mang chỉ số lớn hơn thì thuật toán tìm đường đi ngắn nhất có thể mô tả rất đơn giản:

Gọi $d[v]$ là độ dài đường đi ngắn nhất từ s tới v . Khởi tạo $d[s] = 0$ và $d[v] = +\infty$ với $\forall v \neq s$. Ta sẽ tính các $d[v]$ như sau:

```

for u := 1 to n - 1 do
  for v := u + 1 to n do
    d[v] := min(d[v], d[u] + c[u, v]);

```

(Giả thiết rằng $c[u, v] = +\infty$ nếu như (u, v) không là cung).

Tức là dùng đỉnh u , tối ưu nhãn $d[v]$ của những đỉnh v nối từ u , với u được xét lần lượt từ 1 tới $n - 1$. Có thể làm tốt hơn nữa bằng cách chỉ cần cho u chạy từ đỉnh xuất phát tới đỉnh kết thúc. Bởi lẽ u chạy tới đâu thì nhãn $d[u]$ là không thể cực tiểu hoá thêm nữa.

```

P_4_08_4.PAS * Đường đi ngắn nhất trên đồ thị không có chu trình
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Critical_Path;
const
  InputFile = 'MINPATH.INP';
  OutputFile = 'MINPATH.OUT';
  max = 1000;
  maxEC = 1000;
  maxC = max * maxEC;
var
  c: array[1..max, 1..max] of Integer;
  List, d, Trace: array[1..max] of Integer; {List chứa danh sách các đỉnh theo thứ tự đánh số mới}
  n, s, f: Integer;

procedure LoadGraph; {Nhập dữ liệu}
var
  i, m, u, v: Integer;
  fi: Text;
begin
  Assign(fi, InputFile); Reset(fi);
  ReadLn(fi, n, m, s, f);
  for u := 1 to n do
    for v := 1 to n do
      if u = v then c[u, v] := 0 else c[u, v] := maxC;
  for i := 1 to m do ReadLn(fi, u, v, c[u, v]);

```

```

        Close(fi);
end;

procedure Numbering; {Thuật toán đánh số các đỉnh}
var
    Free: array[1..max] of Boolean;
    u, count: Integer;

procedure Visit(u: Integer);
var
    v: Integer;
begin
    Free[u] := False;
    for v := 1 to n do
        if Free[v] and (c[u, v] <> maxC) then Visit(v);
    List[count] := u;
    Dec(count);
end;

begin
    FillChar(Free, SizeOf(Free), True);
    count := n;
    for u := 1 to n do
        if Free[u] then Visit(u);
end;

procedure Init; {Khởi tạo các nhãn trọng số}
var
    i: Integer;
begin
    for i := 1 to n do d[i] := maxC;
    d[s] := 0;
end;

procedure FindPath; {Thuật toán quy hoạch động tối ưu hoá các d[.]}
var
    i, j, u, v: Integer;
begin
    for i := 1 to n - 1 do
        for j := i + 1 to n do
            begin
                u := List[i]; v := List[j]; {Ánh xạ ngược i, j sang chỉ số cũ u, v}
                if d[v] > d[u] + c[u, v] then
                    begin
                        d[v] := d[u] + c[u, v];
                        Trace[v] := u;
                    end
            end;
end;

procedure PrintResult; {In kết quả}
var
    fo: Text;
begin
    Assign(fo, OutputFile); Rewrite(fo);
    if d[f] = maxC then
        WriteLn(fo, 'There is no path from ', s, ' to ', f)
    else
        begin
            WriteLn(fo, 'Distance from ', s, ' to ', f, ': ', d[f]);
            while f <> s do
                begin
                    Write(fo, f, '<-');

```

```

    f := Trace[f];
  end;
  WriteLn(fo, s);
end;
Close(fo);
end;

begin
  LoadGraph;
  Numbering;
  Init;
  FindPath;
  PrintResult;
end.

```

8.7. ĐƯỜNG ĐI NGẮN NHẤT GIỮA MỌI CẶP ĐỈNH - THUẬT TOÁN FLOYD

Cho đơn đồ thị có hướng, có trọng số $G = (V, E)$ với n đỉnh và m cạnh. Bài toán đặt ra là hãy tính tất cả các $d(u, v)$ là khoảng cách từ u tới v . Rõ ràng là ta có thể áp dụng thuật toán tìm đường đi ngắn nhất xuất phát từ một đỉnh với n khả năng chọn đỉnh xuất phát. Nhưng ta có cách làm gọn hơn nhiều, cách làm này rất giống với thuật toán Warshall mà ta đã biết: Từ ma trận trọng số c , thuật toán Floyd tính lại các $c[u, v]$ thành độ dài đường đi ngắn nhất từ u tới v : Với mọi đỉnh k của đồ thị được xét theo thứ tự từ 1 tới n , xét mọi cặp đỉnh u, v . Cực tiểu hoá $c[u, v]$ theo công thức:

$$c[u, v] := \min(c[u, v], c[u, k] + c[k, v])$$

Tức là nếu như đường đi từ u tới v đang có lại dài hơn đường đi từ u tới k cộng với đường đi từ k tới v thì ta huỷ bỏ đường đi từ u tới v hiện thời và coi đường đi từ u tới v sẽ là nối của hai đường đi từ u tới k rồi từ k tới v (Chú ý rằng ta còn có việc lưu lại vết):

```

for k := 1 to n do
  for u := 1 to n do
    for v := 1 to n do
      c[u, v] := min(c[u, v], c[u, k] + c[k, v]);

```

Tính đúng của thuật toán:

Gọi $c_k[u, v]$ là độ dài đường đi ngắn nhất từ u tới v mà chỉ đi qua các đỉnh trung gian thuộc tập $\{1, 2, \dots, k\}$. Rõ ràng khi $k = 0$ thì $c^0[u, v] = c[u, v]$ (đường đi ngắn nhất là đường đi trực tiếp).

Giả sử ta đã tính được các $c^{k-1}[u, v]$ thì $c^k[u, v]$ sẽ được xây dựng như sau:

Nếu đường đi ngắn nhất từ u tới v mà chỉ qua các đỉnh trung gian thuộc tập $\{1, 2, \dots, k\}$ lại:

❖ Không đi qua đỉnh k thì tức là chỉ qua các đỉnh trung gian thuộc tập $\{1, 2, \dots, k-1\}$ thì

$$c^k[u, v] = c^{k-1}[u, v]$$

❖ Có đi qua đỉnh k thì đường đi đó sẽ là nối của một đường đi từ u tới k và một đường đi từ k tới v , hai đường đi này chỉ đi qua các đỉnh trung gian thuộc tập $\{1, 2, \dots, k-1\}$.

$$c^k[u, v] = c^{k-1}[u, k] + c^{k-1}[k, v].$$

Vì ta muốn $c^k[u, v]$ là cực tiểu nên suy ra: $c^k[u, v] = \min(c^{k-1}[u, v], c^{k-1}[u, k] + c^{k-1}[k, v]).$

Và cuối cùng, ta quan tâm tới $c^n[u, v]$: Độ dài đường đi ngắn nhất từ u tới v mà chỉ đi qua các đỉnh trung gian thuộc tập $\{1, 2, \dots, n\}$.

Khi cài đặt, thì ta sẽ không có các khái niệm $c^k[u, v]$ mà sẽ thao tác trực tiếp trên các trọng số $c[u, v]$. $c[u, v]$ tại bước tối ưu thứ k sẽ được tính toán để tối ưu qua các giá trị $c[u, v]$; $c[u, k]$ và $c[k, v]$ tại bước thứ $k - 1$. Tính chính xác của cách cài đặt dưới dạng ba vòng lặp for lồng như trên có thể thấy được do sự tối ưu bắc cầu chỉ làm tăng tốc độ tối ưu các $c[u, v]$ trong mỗi bước

```
P_4_08_5.PAS * Thuật toán Floyd
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Shortest_Path_by_Floyd;
const
  InputFile = 'MINPATH.INP';
  OutputFile = 'MINPATH.OUT';
  max = 1000;
  maxEC = 1000;
  maxC = max * maxEC;
var
  c: array[1..max, 1..max] of Integer;
  Trace: array[1..max, 1..max] of Integer; {Trace[u, v] = Đỉnh liền sau u trên đường đi từ u tới v}
  n, s, f: Integer;
procedure LoadGraph; {Nhập dữ liệu, đồ thị không được có chu trình âm}
var
  i, m, u, v: Integer;
  fi: Text;
begin
  Assign(fi, InputFile); Reset(fi);
  ReadLn(fi, n, m, s, f);
  for u := 1 to n do
    for v := 1 to n do
      if u = v then c[u, v] := 0 else c[u, v] := maxC;
  for i := 1 to m do ReadLn(fi, u, v, c[u, v]);
  Close(fi);
end;

procedure Floyd;
var
  k, u, v: Integer;
begin
  for u := 1 to n do
    for v := 1 to n do Trace[u, v] := v; {Giả sử đường đi ngắn nhất giữa mọi cặp đỉnh là đường trực tiếp}
  {Thuật toán Floyd}
  for k := 1 to n do
    for u := 1 to n do
      for v := 1 to n do
        if c[u, v] > c[u, k] + c[k, v] then {Đường đi từ qua k tốt hơn}
          begin
            c[u, v] := c[u, k] + c[k, v]; {Ghi nhận đường đi đó thay cho đường cũ}
            Trace[u, v] := Trace[u, k]; {Lưu vết đường đi}
          end;
end;

procedure PrintResult; {In đường đi từ s tới f}
var
  fo: Text;
begin
  Assign(fo, OutputFile); Rewrite(fo);
  if c[s, f] = maxC
```

```

    then WriteLn(fo, 'There is no path from ', s, ' to ', f)
else
begin
  WriteLn(fo, 'Distance from ', s, ' to ', f, ': ', c[s, f]);
repeat
  Write(fo, s, '->');
  s := Trace[s, f]; {Nhắc lại rằng Trace[s, f] là đỉnh liền sau s trên đường đi tới f}
until s = f;
WriteLn(fo, f);
end;
Close(fo);
end;

begin
  LoadGraph;
  Floyd;
  PrintResult;
end.

```

Khác biệt rõ ràng của thuật toán Floyd là khi cần tìm đường đi ngắn nhất giữa một cặp đỉnh khác, chương trình chỉ việc in kết quả chứ không phải thực hiện lại thuật toán Floyd nữa.

8.8. NHẬN XÉT

Bài toán đường đi dài nhất trên đồ thị trong một số trường hợp có thể giải quyết bằng cách đổi dấu trọng số tất cả các cung rồi tìm đường đi ngắn nhất, nhưng hãy cẩn thận, có thể xảy ra trường hợp có chu trình âm.

Trong tất cả các cài đặt trên, vì sử dụng ma trận trọng số chứ không sử dụng danh sách cạnh hay danh sách kè có trọng số, nên ta đều đưa về đồ thị đầy đủ và đếm trọng số $+\infty$ gán cho những cạnh không có trong đồ thị ban đầu. Trên máy tính thì không có khái niệm trừu tượng $+\infty$ nên ta sẽ phải chọn một số dương đủ lớn để thay. Như thế nào là đủ lớn? Số đó phải đủ lớn hơn tất cả trọng số của các đường đi cơ bản để cho dù đường đi thật có tồi tệ đến đâu vẫn tốt hơn đường đi trực tiếp theo cạnh tường tượng ra đó.


Xét về độ phức tạp tính toán, nếu cài đặt như trên, thuật toán Ford-Bellman có độ phức tạp là $O(n^3)$, thuật toán Dijkstra là $O(n^2)$, thuật toán tối ưu nhẫn theo thứ tự tópô là $O(n^2)$ còn thuật toán Floyd là $O(n^3)$. Tuy nhiên nếu sử dụng danh sách kè, thuật toán tối ưu nhẫn theo thứ tự tópô sẽ có độ phức tạp tính toán là $O(m)$. Thuật toán Dijkstra kết hợp với cấu trúc dữ liệu Heap có độ phức tạp $O(\max(n, m).log n)$.

Khác với một bài toán đại số hay hình học có nhiều cách giải thì chỉ cần nắm vững một cách cũng có thể coi là đạt yêu cầu, những thuật toán tìm đường đi ngắn nhất bộc lộ rất rõ ưu, nhược điểm trong từng trường hợp cụ thể (Ví dụ như số đỉnh của đồ thị quá lớn làm cho không thể biểu diễn bằng ma trận trọng số thì thuật toán Floyd sẽ gặp khó khăn, hay thuật toán Ford-Bellman làm việc khá chậm). Vì vậy yêu cầu trước tiên là phải hiểu bản chất và thành thạo trong việc cài đặt tất cả các thuật toán trên để có thể sử dụng chúng một cách uyển chuyển trong từng trường hợp cụ thể. Những bài tập sau đây cho ta thấy rõ điều đó.

Bài tập

Bài 1

Giải thích tại sao đối với đồ thị sau, cần tìm đường đi dài nhất từ đỉnh 1 tới đỉnh 4 lại không thể dùng thuật toán Dijkstra được, cứ thử áp dụng thuật toán Dijkstra theo từng bước xem sao:



Bài 2

Trên mặt phẳng cho n đường tròn ($n \leq 2000$), đường tròn thứ i được cho bởi bộ ba số thực $(x[i], y[i], R[i])$, $(x[i], y[i])$ là toạ độ tâm và $R[i]$ là bán kính. Chi phí di chuyển trên mỗi đường tròn bằng 0. Chi phí di chuyển giữa hai đường tròn bằng khoảng cách giữa chúng. Hãy tìm phương án di chuyển giữa hai đường tròn s, f cho trước với chi phí ít nhất.

Bài 3

Cho một dãy n số nguyên $a[1..n]$ ($n \leq 10000$; $1 \leq a[i] \leq 10000$). Hãy tìm một dãy con gồm nhiều nhất các phần tử của dãy đã cho mà tổng của hai phần tử liên tiếp là số nguyên tố.

Hướng dẫn: Dựng đồ thị $G = (V, E)$, $V = \{1, \dots, n\}$, $(i, j) \in E$ nếu $i < j$ và $a[i] + a[j]$ là số nguyên tố, tìm đường đi dài nhất trên đồ thị G không có chu trình

Bài 4

Một công trình lớn được chia làm n công đoạn đánh số 1, 2, ..., n . Công đoạn i phải thực hiện mất thời gian $t[i]$. Quan hệ giữa các công đoạn được cho bởi bảng $a[i, j]$: $a[i, j] = \text{TRUE} \Leftrightarrow$ công đoạn j chỉ được bắt đầu khi mà công việc i đã xong. Hai công đoạn độc lập nhau có thể tiến hành song song, hãy bố trí lịch thực hiện các công đoạn sao cho thời gian hoàn thành cả công trình là sớm nhất, cho biết thời gian sớm nhất đó.

Hướng dẫn: Dựng đồ thị $G = (V, E)$, trong đó:

$$V = \{0, 1, \dots, n\}$$

$$E = \{(0, u) | 1 \leq u \leq n\} \cup \{(u, v) | a[u, v] = \text{TRUE}\}. \text{ Trọng số mỗi cung } (i, j) \text{ đặt bằng } t[j]$$

Tìm đường đi dài nhất trên đồ thị không có chu trình, thời điểm thích hợp để thực hiện công việc i chính là $d[i] - t[i]$.

Bài 5

Cho một bảng các số tự nhiên kích thước $m \times n$ ($1 \leq m, n \leq 100$). Từ một ô có thể di chuyển sang một ô kề cạnh với nó. Hãy tìm một cách đi từ ô (x, y) ra một ô biên sao cho tổng các số ghi trên các ô đi qua là cực tiểu.

Hướng dẫn: Dùng thuật toán Dijkstra với cấu trúc Heap

Bài 6: Arbitrage

Arbitrage là một cách sử dụng sự bất hợp lý trong hối đoái tiền tệ để kiếm lời.

Ví dụ:

Giả sử

1\$ mua được 0.7£

1£ mua được 190¥

1¥ mua được 0.009\$

Từ 1\$, ta có thể đổi sang 0.7£, sau đó sang $0.7 \times 190 = 133$ ¥, rồi đổi lại sang $133 \times 0.009 = 1.197$ \$.
Kiếm được 0.197\$ lãi.

Giả sử rằng có n loại tiền tệ đánh số 1, 2, ..., n. Và một bảng R kích thước nxn cho biết tỉ lệ hối đoái. Một đơn vị tiền i đổi được $R[i, j]$ đơn vị tiền j.

a) Hãy tìm thuật toán để xác định xem có thể kiếm lời từ bảng tỉ giá hối đoái này bằng phương pháp Arbitrage hay không?

b) Giả sử rằng nhà băng đủ thông minh để vô hiệu hóa Arbitrage, tính tỉ lệ tốt nhất có thể hoán đổi giữa hai loại tiền tệ bất kỳ.

Hướng dẫn:

Đặt $R[i, j] := -\ln(R[i, j])$. Dựng đồ thị $G = (V, E)$ có n đỉnh và ma trận trọng số là R. Thêm vào G một đỉnh đánh số 0 và nối nó tới tất cả những đỉnh còn lại bằng cung có trọng số 0.

a) Dùng thuật toán Ford-Bellman tìm đường đi ngắn nhất xuất phát từ 0. Sau khi thuật toán Ford-Bellman kết thúc, xét tất cả các cặp đỉnh. Nếu tồn tại một cặp (u, v) mà $d[v] > d[u] + R[u, v]$ thì thông báo tồn tại Arbitrage. Thuật toán này dựa trên nhận xét: Tỉ giá hối đoái đã cho có thể dùng Arbitrage nếu và chỉ nếu đồ thị có chu trình âm.

b) Nếu tỉ giá hối đoái không cho phép Arbitrage tức là đồ thị G không có chu trình âm. Dùng thuật toán Floyd tìm đường đi ngắn nhất giữa mọi cặp đỉnh. Tỉ lệ tốt nhất có thể hoán đổi từ loại tiền i sang loại tiền j chính là $e^{-d[i, j]}$, trong đó $d[i, j]$ là độ dài đường đi ngắn nhất giữa hai đỉnh i và j trên G.

Bài 7:

Cho một đồ thị $G = (V, E)$ gồm các cạnh được gán trọng số không âm. Cho hai đỉnh A và B. Hãy chỉ ra hai đường đi từ A tới B thỏa mãn:

- ❖ Hai đường đi này không có cạnh chung
- ❖ Tổng độ dài hai đường đi là nhỏ nhất có thể

Hướng dẫn:

Coi mỗi cạnh của đồ thị tương đương với hai cung có hướng ngược chiều nhau, trọng số trên cung (u, v) được gán bằng $c[u, v]$.


Dùng thuật toán Dijkstra tìm đường đi ngắn nhất từ A tới B: $\langle v_0, v_1, \dots, v_p \rangle$. Dọc trên đường đi Dijkstra, với mỗi cạnh (v_{i-1}, v_i) , ta bỏ đi cung (v_{i-1}, v_i) , giữ lại cung (v_i, v_{i-1}) và gán trọng số cung này $= -c[v_i, v_{i-1}]$). Sau phép biến đổi có những cung trọng số âm nhưng không tạo thành chu trình âm.

Dùng thuật toán Ford-Bellman tìm đường đi ngắn nhất từ A tới B: $\langle u_0, u_1, \dots, u_q \rangle$. Dọc trên đường đi Ford-Bellman, với mỗi cạnh (u_{i-1}, u_i) , ta bỏ đi cung (u_{i-1}, u_i) .


Với mỗi cung (u, v) của đồ thị, nếu cả cung (u, v) và (v, u) đều được duy trì đến bước này thì loại bỏ luôn cả hai. Những cung còn lại chỉ ra hai đường đi từ B về A, bằng cách lật ngược chiều đường đi, ta có lời giải.

Ví dụ:

Đồ thị ban đầu:




Đường đi Dijkstra $\langle 1, 2, 4, 5 \rangle$ (Độ dài 10):




Bỏ đi các cung $(1, 2), (2, 4), (4, 5)$. Đặt lại trọng số các cung ngược chiều đường đi:


$$c[2, 1] := -3; c[4, 2] := -4; c[5, 4] := -3;$$




Đường đi Ford-Bellman $\langle 1, 4, 2, 3, 5 \rangle$ (Độ dài 14):



Bỏ đi các cung $(1, 4), (4, 2), (2, 3), (3, 5)$:



Bỏ nốt cung $(3, 4)$ và $(4, 3)$ ta có 2 đường đi từ 5 về 1: $\langle 5, 4, 1 \rangle$ và $\langle 5, 3, 2, 1 \rangle$.



Đối chiếu lên đồ thị ban đầu, tổng độ dài 2 đường đi tìm được = Độ dài đường đi Dijkstra + Độ dài đường đi Ford-Bellman = 24. Lật ngược thứ tự các đỉnh trong hai đường đi trên, ta được cặp đường đi từ 1 tới 5 có tổng độ dài nhỏ nhất cần tìm.

§9. BÀI TOÁN CÂY KHUNG NHỎ NHẤT


9.1. BÀI TOÁN CÂY KHUNG NHỎ NHẤT

Cho $G = (V, E)$ là đồ thị vô hướng liên thông có trọng số, với một cây khung T của G , ta gọi trọng số của cây T là tổng trọng số các cạnh trong T . Bài toán đặt ra là trong số các cây khung của G , chỉ ra cây khung có trọng số nhỏ nhất, cây khung như vậy được gọi là cây khung (hay cây bao trùm) nhỏ nhất của đồ thị, và bài toán đó gọi là bài toán cây khung nhỏ nhất. Sau đây ta sẽ xét hai thuật toán thông dụng để giải bài toán cây khung nhỏ nhất của đơn đồ thị vô hướng có trọng số. Bạn có thể đưa vào một số sửa đổi nhỏ trong thủ tục nhập liệu để giải quyết bài toán trong trường hợp đa đồ thị.

Input: file văn bản MINTREE.INP:

- ❖ Dòng 1: Ghi hai số số đỉnh n (≤ 1000) và số cạnh m của đồ thị cách nhau ít nhất 1 dấu cách
- ❖ m dòng tiếp theo, mỗi dòng có dạng 3 số $u, v, c[u, v]$ cách nhau ít nhất 1 dấu cách thể hiện đồ thị có cạnh (u, v) và trọng số cạnh đó là $c[u, v]$. ($c[u, v]$ là số nguyên có giá trị tuyệt đối không quá 1000).

Output: file văn bản MINTREE.OUT ghi các cạnh thuộc cây khung và trọng số cây khung



MINTREE.INP	MINTREE.OUT
6 9	Minimal spanning tree:
1 2 1	$(2, 4) = 1$
1 3 1	$(3, 6) = 1$
2 4 1	$(2, 5) = 1$
2 3 2	$(1, 3) = 1$
2 5 1	$(1, 2) = 1$
3 5 1	Weight = 5
3 6 1	
4 5 2	
5 6 2	

9.2. THUẬT TOÁN KRUSKAL (JOSEPH KRUSKAL - 1956)

Thuật toán Kruskal dựa trên mô hình xây dựng cây khung bằng thuật toán hợp nhất (§5), chỉ có điều thuật toán không phải xét các cạnh với thứ tự tự ý mà xét các cạnh theo thứ tự đã sắp xếp: Với đồ thị vô hướng $G = (V, E)$ có n đỉnh. Khởi tạo cây T ban đầu không có cạnh nào. Xét tất cả các cạnh của đồ thị từ cạnh có trọng số nhỏ đến cạnh có trọng số lớn, nếu việc thêm cạnh đó vào T không tạo thành chu trình đơn trong T thì kết nạp thêm cạnh đó vào T . Cứ làm như vậy cho tới khi:

- ❖ Hoặc đã kết nạp được $n - 1$ cạnh vào trong T thì ta được T là cây khung nhỏ nhất
- ❖ Hoặc chưa kết nạp đủ $n - 1$ cạnh nhưng hễ cứ kết nạp thêm một cạnh bất kỳ trong số các cạnh còn lại thì sẽ tạo thành chu trình đơn. Trong trường hợp này đồ thị G là không liên thông, việc tìm kiếm cây khung thất bại.

Như vậy có hai vấn đề quan trọng khi cài đặt thuật toán Kruskal:

Thứ nhất, làm thế nào để xét được các cạnh từ cạnh có trọng số nhỏ tới cạnh có trọng số lớn. Ta có thể thực hiện bằng cách sắp xếp danh sách cạnh theo thứ tự không giảm của trọng số, sau đó duyệt từ đầu tới cuối danh sách cạnh. Trong trường hợp tổng quát, thuật toán HeapSort là hiệu quả nhất bởi nó cho phép chọn lần lượt các cạnh từ cạnh trọng số nhỏ nhất tới cạnh trọng số lớn nhất ra khỏi Heap và có thể xử lý (bỏ qua hay thêm vào cây) luôn.

Thứ hai, làm thế nào kiểm tra xem việc thêm một cạnh có tạo thành chu trình đơn trong T hay không. Để ý rằng các cạnh trong T ở các bước sẽ tạo thành một rừng (đó thị không có chu trình đơn). Muốn thêm một cạnh (u, v) vào T mà không tạo thành chu trình đơn thì (u, v) phải nối hai cây khác nhau của rừng T, bởi nếu u, v thuộc cùng một cây thì sẽ tạo thành chu trình đơn trong cây đó. Ban đầu, ta khởi tạo rừng T gồm n cây, mỗi cây chỉ gồm đúng một đỉnh, sau đó, mỗi khi xét đến **cạnh nối hai cây khác nhau** của rừng T thì ta **kết nạp cạnh đó** vào T, đồng thời **hợp nhất hai cây đó lại thành một cây**.


Nếu cho mỗi đỉnh v trên cây một nhãn $\text{Lab}[v]$ là số hiệu đỉnh cha của đỉnh v trong cây, trong trường hợp v là gốc của một cây thì $\text{Lab}[v]$ được gán một giá trị âm. Khi đó ta hoàn toàn có thể xác định được gốc của cây chứa đỉnh v bằng hàm `GetRoot` dưới đây:

```
function GetRoot(v∈V): ∈V;
begin
  while Lab[v] > 0 do v := Lab[v];
  GetRoot := v;
end;
```

Vậy để kiểm tra một cạnh (u, v) có nối hai cây khác nhau của rừng T hay không? ta có thể kiểm tra $\text{GetRoot}(u)$ có khác $\text{GetRoot}(v)$ hay không, bởi mỗi cây chỉ có duy nhất một gốc.

Để hợp nhất cây gốc r_1 và cây gốc r_2 thành một cây, ta lưu ý rằng mỗi cây ở đây chỉ dùng để ghi nhận một tập hợp đỉnh thuộc cây đó chứ cấu trúc cạnh trên cây thế nào thì không quan trọng. Vậy để hợp nhất cây gốc r_1 và cây gốc r_2 , ta chỉ việc coi r_1 là nút cha của r_2 trong cây bằng cách đặt:

$$\text{Lab}[r_2] := r_1.$$



Hình 76: Hai cây gốc r_1 và r_2 và cây mới khi hợp nhất chúng

Tuy nhiên, để thuật toán làm việc hiệu quả, tránh trường hợp cây tạo thành bị suy biến khiến cho hàm `GetRoot` hoạt động chậm, người ta thường đánh giá: Để hợp hai cây lại thành một, thì gốc cây nào ít nút hơn sẽ coi là con của gốc cây kia.

Thuật toán hợp nhất cây gốc r_1 và cây gốc r_2 có thể viết như sau:

```
{Count[k] là số đỉnh của cây gốc k}
procedure Union(r1, r2 ∈ V);
begin
  if Count[r1] < Count[r2] then {Hợp nhất thành cây gốc r2}
    begin
      Count[r2] := Count[r1] + Count[r2];
      Lab[r1] := r2;
    end
  else {Hợp nhất thành cây gốc r1}
    begin
      Count[r1] := Count[r1] + Count[r2];
      Lab[r2] := r1;
    end;
end;
```

Khi cài đặt, ta có thể tận dụng ngay nhãn Lab[r] để lưu số đỉnh của cây gốc r, bởi nhu đã giải thích ở trên, Lab[r] chỉ cần mang một giá trị âm là đủ, vậy ta có thể coi Lab[r] = -Count[r] với r là gốc của một cây nào đó.

```
procedure Union(r1, r2 ∈ V); {Hợp nhất cây gốc r1 với cây gốc r2}
begin
  x := Lab[r1] + Lab[r2]; {-x là tổng số nút của cả hai cây}
  if Lab[r1] > Lab[r2] then {Cây gốc r1 ít nút hơn cây gốc r2, hợp nhất thành cây gốc r2}
    begin
      Lab[r1] := r2; {r2 là cha của r1}
      Lab[r2] := x; {r2 là gốc cây mới, -Lab[r2] giờ đây là số nút trong cây mới}
    end
  else {Hợp nhất thành cây gốc r1}
    begin
      Lab[r1] := x; {r1 là gốc cây mới, -Lab[r1] giờ đây là số nút trong cây mới}
      Lab[r2] := r1; {cha của r2 sẽ là r1}
    end;
end;
```

Mô hình thuật toán Kruskal:

```
for ∀k ∈ V do Lab[k] := -1;
for (∀(u, v) ∈ E theo thứ tự từ cạnh trọng số nhỏ tới cạnh trọng số lớn) do
begin
  r1 := GetRoot(u); r2 := GetRoot(v);
  if r1 ≠ r2 then {(u, v) nối hai cây khác nhau}
    begin
      <Kết nạp (u, v) vào cây, nếu đã đủ n - 1 cạnh thì thuật toán dừng>
      Union(r1, r2); {Hợp nhất hai cây lại thành một cây}
    end;
end;
```

P_4_09_1.PAS * Thuật toán Kruskal

```
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Minimal_Spanning_Tree_by_Kruskal;
const
  InputFile = 'MINTREE.INP';
  OutputFile = 'MINTREE.OUT';
  maxV = 1000;
  maxE = (maxV - 1) * maxV div 2;
type
  TEdge = record {Cấu trúc một cạnh}
    u, v, c: Integer; {Hai đỉnh và trọng số}
    Mark: Boolean; {Đánh dấu có được kết nạp vào cây khung hay không}
  end;
var
  e: array[1..maxE] of TEdge; {Danh sách cạnh}
```

```

Lab: array[1..maxV] of Integer; {Lab[v] là đỉnh cha của v, nếu v là gốc thì Lab[v] = - số con cây gốc v}
n, m: Integer;
Connected: Boolean;

procedure LoadGraph;
var
  i: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n, m);
  for i := 1 to m do
    with e[i] do
      ReadLn(f, u, v, c);
  Close(f);
end;

procedure Init;
var
  i: Integer;
begin
  for i := 1 to n do Lab[i] := -1; {Rừng ban đầu, mọi đỉnh là gốc của cây gồm đúng một nút}
  for i := 1 to m do e[i].Mark := False;
end;

function GetRoot(v: Integer): Integer; {Lấy gốc của cây chứa v}
begin
  while Lab[v] > 0 do v := Lab[v];
  GetRoot := v;
end;

procedure Union(r1, r2: Integer); {Hợp nhất hai cây lại thành một cây}
var
  x: Integer;
begin
  x := Lab[r1] + Lab[r2];
  if Lab[r1] > Lab[r2] then
    begin
      Lab[r1] := r2;
      Lab[r2] := x;
    end
  else
    begin
      Lab[r1] := x;
      Lab[r2] := r1;
    end;
end;

procedure AdjustHeap(root, last: Integer); {Vun thành đóng, dùng cho HeapSort}
var
  Key: TEdge;
  child: Integer;
begin
  Key := e[root];
  while root * 2 <= Last do
    begin
      child := root * 2;
      if (child < Last) and (e[child + 1].c < e[child].c)
        then Inc(child);
      if Key.c <= e[child].c then Break;
      e[root] := e[child];
      root := child;
    end;
end;

```

```

e[root] := Key;
end;

procedure Kruskal;
var
  i, r1, r2, Count, a: Integer;
  tmp: TEdge;
begin
  Count := 0;
  Connected := False;
  for i := m div 2 downto 1 do AdjustHeap(i, m); {Vun danh sách cạnh thành đồng}
  for i := m - 1 downto 0 do {Rút lần lượt các cạnh khỏi đồng, từ cạnh ngắn tới cạnh dài}
    begin
      tmp := e[1]; e[1] := e[i + 1]; e[i + 1] := tmp;
      AdjustHeap(1, i);
      r1 := GetRoot(e[i + 1].u); r2 := GetRoot(e[i + 1].v);
      if r1 <> r2 then {Cạnh e[i + 1] nối hai cây khác nhau}
        begin
          e[i + 1].Mark := True; {Kết nạp cạnh đó vào cây}
          Inc(Count); {Đếm số cạnh}
          if Count = n - 1 then {Nếu đã đủ số thì thành công}
            begin
              Connected := True;
              Exit;
            end;
          Union(r1, r2); {Hợp nhất hai cây thành một cây}
        end;
    end;
  end;
end;

procedure PrintResult;
var
  i, Count, W: Integer;
  f: Text;
begin
  Assign(f, OutputFile); Rewrite(f);
  if not Connected then
    WriteLn(f, 'Error: Graph is not connected')
  else
    begin
      WriteLn(f, 'Minimal spanning tree: ');
      Count := 0;
      W := 0;
      for i := 1 to m do {Duyệt danh sách cạnh}
        with e[i] do
          begin
            if Mark then {Lọc ra những cạnh đã kết nạp vào cây khung}
              begin
                WriteLn(f, '(', u, ', ', v, ') = ', c);
                Inc(Count);
                W := W + c;
              end;
            if Count = n - 1 then Break; {Cho tới khi đủ n - 1 cạnh}
          end;
      WriteLn(f, 'Weight = ', W);
    end;
  Close(f);
end;

begin
  LoadGraph;
  Init;
  Kruskal;

```

```

PrintResult;
end.
```

Xét về độ phức tạp tính toán, ta có thể chứng minh được rằng thao tác GetRoot có độ phức tạp là $O(\lg n)$, còn thao tác Union là $O(1)$. Giả sử ta đã có danh sách cạnh đã sắp xếp rồi thì xét vòng lặp dựng cây khung, nó duyệt qua danh sách cạnh và với mỗi cạnh nó gọi 2 lần thao tác GetRoot, vậy thì độ phức tạp là $O(m\lg n)$, nếu đồ thị có cây khung thì $m \geq n-1$ nên ta thấy chi phí thời gian chủ yếu sẽ nằm ở thao tác sắp xếp danh sách cạnh bởi độ phức tạp của HeapSort là $O(m\lg m)$. Vậy độ phức tạp tính toán của thuật toán là $O(m\lg m)$ trong trường hợp xấu nhất. Tuy nhiên, phải lưu ý rằng để xây dựng cây khung thì ít khi thuật toán phải duyệt toàn bộ danh sách cạnh mà chỉ một phần của danh sách cạnh mà thôi.

9.3. THUẬT TOÁN PRIM (ROBERT PRIM - 1957)

Thuật toán Kruskal hoạt động chậm trong trường hợp đồ thị dày (có nhiều cạnh). Trong trường hợp đó người ta thường sử dụng phương pháp lân cận gần nhất của Prim. Thuật toán đó có thể phát biểu hình thức như sau:

Đơn đồ thị vô hướng $G = (V, E)$ có n đỉnh được cho bởi ma trận trọng số C . Qui ước $c[u, v] = +\infty$ nếu (u, v) không là cạnh. Xét cây T trong G và một đỉnh v , gọi **khoảng cách từ v tới T** là trọng số nhỏ nhất trong số các cạnh nối v với một đỉnh nào đó trong T :

$$d[v] = \min \{c[u, v] \mid u \in T\}$$

Ban đầu khởi tạo cây T chỉ gồm có mỗi đỉnh $\{1\}$. Sau đó cứ chọn trong số các đỉnh ngoài T ra một đỉnh gần T nhất, kết nạp đỉnh đó vào T đồng thời kết nạp luôn cả cạnh tạo ra khoảng cách gần nhất đó. Cứ làm như vậy cho tới khi:

- ❖ Hoặc đã kết nạp được tất cả n đỉnh thì ta có T là cây khung nhỏ nhất
- ❖ Hoặc chưa kết nạp được hết n đỉnh nhưng mọi đỉnh ngoài T đều có khoảng cách tới T là $+\infty$. Khi đó đồ thị đã cho không liên thông, ta thông báo việc tìm cây khung thất bại.

Về mặt kỹ thuật cài đặt, ta có thể bắt đầu từ một cây rỗng và khởi tạo $d[1] := 0$ còn $d[v] := +\infty$ với $\forall v \neq 1$. Tại mỗi bước chọn đỉnh đưa vào T , ta sẽ chọn đỉnh u nào ngoài T và có $d[u]$ nhỏ nhất. Khi kết nạp u vào T rồi thì các nhãn $d[v]$ sẽ thay đổi: $d[v]_{\text{mới}} := \min(d[v]_{\text{ cũ}}, c[u, v])$. Bước lặp đầu tiên sẽ kết nạp đỉnh 1 vào cây, từ bước lặp thứ hai, trước khi kết nạp một đỉnh vào cây, ta lưu lại cạnh tạo ra khoảng cách gần nhất từ cây tới đỉnh đó để cuối cùng in ra cây khung nhỏ nhất.

```

for (forall v in V) do
begin
  d[v] := +infinity;
  Free[v] := True; {Chưa có đỉnh nào ∈ cây}
end;
d[1] := 0;
for k := 1 to n do
begin
  u := arg min(d[v] | v ∈ V and Free[v] = True); {Chọn u là có nhãn tự do nhỏ nhất}
  Free[u] := False; {Cô đính nhãn đỉnh u ⇔ kết nạp u vào cây}
  for (forall v ∈ V: (u, v) ∈ E) do
    if d[v] > c[u, v] then
      d[v] := c[u, v];

```

```

begin
  d[v] := c[u, v];
  Trace[v] := u;
end;
end;
{Thông báo cây khung gồm có các cạnh (Trace[v], v) với  $\forall v \in V: v \neq 1$ };

P_4_09_2.PAS * Thuật toán Prim
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Minimal_Spanning_Tree_by_Prim;
const
  InputFile = 'MINTREE.INP';
  OutputFile = 'MINTREE.OUT';
  max = 1000;
  maxCE = 1000;
  maxC = max * maxCE;
var
  c: array[1..max, 1..max] of Integer;
  d: array[1..max] of Integer;
  Free: array[1..max] of Boolean;
  Trace: array[1..max] of Integer; {Vết, Trace[v] là đỉnh cha của v trong cây khung nhỏ nhất}
  n, m: Integer;
  Connected: Boolean;

procedure LoadGraph; {Nhập đồ thị}
var
  i, u, v: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, n, m);
  for u := 1 to n do
    for v := 1 to n do
      if u = v then c[u, v] := 0 else c[u, v] := maxC; {Khởi tạo ma trận trọng số}
  for i := 1 to m do
    begin
      ReadLn(f, u, v, c[u, v]);
      c[v, u] := c[u, v];
    end;
  Close(f);
end;

procedure Init;
var
  v: Integer;
begin
  v := 1; {Định 1 có nhãn khoảng cách là 0}
  for v := 2 to n do d[v] := maxC; {Các đỉnh khác có nhãn khoảng cách +∞}
  FillChar(Free, SizeOf(Free), True); {Cây T ban đầu là rỗng}
end;

procedure Prim;
var
  k, i, u, v, min: Integer;
begin
  Connected := True;
  for k := 1 to n do
    begin
      u := 0; min := maxC; {Chọn đỉnh u chưa bị kết nạp có d[u] nhỏ nhất}
      for i := 1 to n do
        if Free[i] and (d[i] < min) then
          begin
            min := d[i];
            v := i;
          end;
      if v > 0 then
        begin
          Trace[v] := u;
          d[v] := min;
          Free[v] := False;
        end;
    end;
end;

```

```

        u := i;
      end;
    if u = 0 then {Nếu không chọn được u nào có d[u] < +∞ thì đồ thị không liên thông}
      begin
        Connected := False;
        Break;
      end;
    Free[u] := False; {Nếu chọn được thì đánh dấu u đã bị kết nạp, lặp lần 1 thì d[i] nhiên u = 1 bởi d[1] = 0}
    for v := 1 to n do
      if Free[v] and (d[v] > c[u, v]) then {Tính lại các nhãn khoảng cách d[v] (v chưa kết nạp)}
        begin
          d[v] := c[u, v]; {Tôi ưu nhãnh d[v] theo công thức}
          Trace[v] := u; {Lưu vết, định nối với v cho khoảng cách ngắn nhất là u}
        end;
      end;
    end;

procedure PrintResult;
var
  v, W: Integer;
  f: Text;
begin
  Assign(f, OutputFile); Rewrite(f);
  if not Connected then {Nếu đồ thị không liên thông thì thất bại}
    WriteLn(f, 'Error: Graph is not connected')
  else
    begin
      WriteLn(f, 'Minimal spanning tree: ');
      W := 0;
      for v := 2 to n do {Cây khung nhỏ nhất gồm những cạnh (v, Trace[v])}
        begin
          WriteLn(f, '(', Trace[v], ', ', v, ') = ', c[Trace[v], v]);
          W := W + c[Trace[v], v];
        end;
      WriteLn(f, 'Weight = ', W);
    end;
  Close(f);
end;

begin
  LoadGraph;
  Init;
  Prim;
  PrintResult;
end.

```

Xét về độ phức tạp tính toán, thuật toán Prim có độ phức tạp là $O(n^2)$. Tương tự thuật toán Dijkstra, nếu kết hợp thuật toán Prim với cấu trúc Heap sẽ được một thuật toán với độ phức tạp $O((m+n)lgn)$. Tuy nhiên nếu phải làm việc với đồ thị thưa, người ta thường sử dụng thuật toán Kruskal để tìm cây khung chứ không dùng thuật toán Prim với cấu trúc Heap.

Bài tập

Bài 1

So sánh hiệu quả của thuật toán Kruskal và thuật toán Prim về tốc độ.

Bài 2

Trên một nền phẳng với hệ toạ độ Decartes vuông góc đặt n máy tính, máy tính thứ i được đặt ở toạ độ $(x[i], y[i])$. Đã có sẵn một số dây cáp mạng nối giữa một số cặp máy tính. Cho phép

nối thêm các dây cáp mạng nối giữa từng cặp máy tính. Chi phí nối một dây cáp mạng tỉ lệ thuận với khoảng cách giữa hai máy cần nối. Hãy tìm cách nối thêm các dây cáp mạng để cho các máy tính trong toàn mạng là liên thông và chi phí nối mạng là nhỏ nhất.

Hướng dẫn:

Xây dựng đồ thị đầy đủ $G = (V, E)$. Mỗi đỉnh tương ứng với một máy tính.

Trọng số cạnh (u, v) sẽ được đặt bằng

❖ 0, nếu đã có sẵn cáp mạng nối hai máy u, v .

❖ $\sqrt{(x_u - x_v)^2 + (y_u - y_v)^2}$, nếu chưa có sẵn cáp mạng nối hai máy u và v

Tìm cây khung nhỏ nhất của G . Những cạnh trọng số $\neq 0$ tương ứng với những cáp mạng cần lắp đặt thêm.

Bài 3

Hệ thống điện trong thành phố được cho bởi n trạm biến thế và các đường dây điện nối giữa các cặp trạm biến thế. Mỗi đường dây điện e có độ an toàn là $p(e)$. Ở đây $0 < p(e) \leq 1$. Độ an toàn của cả lưới điện là tích độ an toàn trên các đường dây. Ví dụ như có một đường dây nguy hiểm: $p(e) = 1\%$ thì cho dù các đường dây khác là tuyệt đối an toàn (độ an toàn = 100%) thì độ an toàn của mạng cũng rất thấp (1%). Hãy tìm cách bỏ đi một số dây điện để cho các trạm biến thế vẫn liên thông và độ an toàn của mạng là lớn nhất có thể.

Hướng dẫn:

Bằng kỹ thuật lấy logarithm, độ an toàn trên lưới điện trở thành tổng độ an toàn trên các đường dây. Dựng đồ thị có mỗi đỉnh tương ứng với một trạm biến thế và mỗi cạnh tương ứng với một đường dây điện e được gán trọng số là $-\ln(p(e))$. Tìm cây khung nhỏ nhất rồi loại bỏ tất cả những đường dây điện tương ứng với các cạnh không nằm trên cây khung.

Bài 4

Hãy thử cài đặt thuật toán Prim với cấu trúc dữ liệu Heap chứa các đỉnh ngoài cây, tương tự như đối với thuật toán Dijkstra.

§10. BÀI TOÁN LUỒNG CỰC ĐẠI TRÊN MẠNG

10.1. CÁC KHÁI NIỆM

10.1.1. Mạng

Ta gọi mạng (network) là một đồ thị có hướng $G = (V, E)$ gồm n đỉnh và m cung, trong đó có hai đỉnh phân biệt s và t , đỉnh s gọi là điểm phát (source) và đỉnh t gọi là đỉnh thu (sink). Mỗi cung $e = (u, v) \in E$ được gán với một số không âm $c(e) = c[u, v]$ gọi là khả năng thông qua của cung đó (capacity). Để thuận tiện cho việc trình bày, ta qui ước rằng nếu mạng không có cung (u, v) thì ta thêm vào cung (u, v) với khả năng thông qua $c[u, v]$ bằng 0.

10.1.2. Định nghĩa 1


Nếu có mạng $G = (V, E)$. Ta gọi luồng Φ trong mạng G là một phép gán cho mỗi cung $e = (u, v)$ một số thực không âm $\Phi(e) = \Phi[u, v]$ gọi là luồng trên cung e , thoả mãn 2 tính chất:

❖ Tính chất 1: (Capacity constraint): Luồng trên mỗi cung không được vượt quá khả năng thông qua của cung đó: $0 \leq \Phi[u, v] \leq c[u, v]$ (Với $\forall u, v \in V$).

❖ Tính chất 2 (Flow conservation): Với mỗi đỉnh v không phải đỉnh phát và cũng không phải đỉnh thu, tổng luồng trên các cung đi vào v bằng tổng luồng trên các cung đi ra khỏi v :

$$\sum_{u \in V} \Phi[u, v] = \sum_{w \in V} \Phi[v, w] \quad (\text{Với } \forall v \in V \setminus \{s, t\})$$

Giá trị của một luồng được định nghĩa bằng: tổng luồng trên các cung đi ra khỏi đỉnh phát trừ đi tổng luồng trên các cung đi vào đỉnh phát: $|\Phi| = \sum_{u \in V} \Phi[s, u] - \sum_{v \in V} \Phi[v, s]$.



Hình 77: Mạng với các khả năng thông qua (1 phát, 6 thu) và một luồng của nó với giá trị 7

Bởi giá trị của luồng Φ trên các cung là số không âm, đôi khi người ta còn gọi Φ là luồng dương (positive flow) trên mạng G . Một số tài liệu khác đưa vào thêm ràng buộc đỉnh s không được có cung đi vào và đỉnh t không có cung đi ra, khi đó giá trị luồng được tính bằng tổng luồng trên các cung đi ra khỏi đỉnh phát. Cách hiểu này có thể quy về một trường hợp riêng của định nghĩa.

10.1.3. Bài toán luồng cực đại trên mạng

Cho một mạng G , hãy tìm luồng Φ^* có giá trị lớn nhất.

Không giảm tính tổng quát, trong bài toán tìm luồng cực đại, ta có thể giả thiết rằng với mọi luồng Φ thì luồng trên cung (u, v) và luồng trên cung (v, u) không đồng thời là số dương ($\forall u, v \in V$). Bởi nếu không ta chỉ việc bớt cả $\Phi[u, v]$ và $\Phi[v, u]$ đi một lượng bằng $\min(\Phi[u, v], \Phi[v, u])$ thì được một luồng mới có giá trị bằng luồng ban đầu trong đó hoặc $\Phi[u, v]$ bằng 0 hoặc $\Phi[v, u]$ bằng 0.

10.1.4. Định nghĩa 2

Nếu có mạng $G = (V, E)$, ta gọi luồng f trên mạng G là một phép gán cho mỗi cung $e = (u, v)$ một số thực $f(e) = f[u, v]$ gọi là luồng trên cung e , thoả mãn 3 tính chất:

- ❖ Tính chất 1: (Capacity constraint): Luồng trên mỗi cung không được vượt quá khả năng thông qua của cung đó: $f[u, v] \leq c[u, v]$ với $\forall u, v \in V$.
- ❖ Tính chất 2: (Skew symmetry): Với $\forall u, v \in V$, luồng trên cung (u, v) và luồng trên cung (v, u) có cùng giá trị tuyệt đối nhưng trái dấu nhau: $f[u, v] = -f[v, u]$.
- ❖ Tính chất 3: (Flow conservation): Với mỗi đỉnh u không phải đỉnh phát và cũng không phải đỉnh thu, tổng luồng trên các cung đi ra khỏi u bằng 0: $\sum_{v \in V} f[u, v] = 0$

Giá trị của một luồng được định nghĩa bằng: tổng luồng trên các cung đi ra khỏi đỉnh phát:

$$|f| = \sum_{u \in V} f[s, u].$$

Định lý 1: Hai định nghĩa 1 và 2 là tương đương.

Chứng minh:

Từ “tương đương” ở đây hiểu theo mục đích của bài toán tìm luồng cực đại, có nghĩa là việc tìm luồng dương cực đại theo định nghĩa 1 tương đương với việc tìm luồng cực đại theo định nghĩa 2.

(1) \Rightarrow (2): Nếu có một luồng dương Φ trên G theo định nghĩa 1, ta xây dựng luồng f như sau:

$$f[u, v] := \Phi[u, v] - \Phi[v, u]$$

Ta chứng minh f thoả mãn ba điều kiện của luồng theo định nghĩa 2:

- ❖ Tính chất 1: $\forall u, v \in V$, do $\Phi[v, u] \geq 0$ nên $f[u, v] = \Phi[u, v] - \Phi[v, u] \leq c[u, v]$
- ❖ Tính chất 2: $\forall u, v \in V$, ta có $f[u, v] = \Phi[u, v] - \Phi[v, u] = -(\Phi[v, u] - \Phi[u, v]) = -f[v, u]$
- ❖ Tính chất 3: $\forall u \in V$, ta có $\sum_{v \in V} f[u, v] = \sum_{v \in V} (\Phi[u, v] - \Phi[v, u]) = \sum_{v \in V} \Phi[u, v] - \sum_{v \in V} \Phi[v, u] = 0$

Về giá trị luồng, ta có:

$$|f| = \sum_{v \in V} f[s, v] = \sum_{v \in V} (\Phi[s, v] - \Phi[v, s]) = \sum_{v \in V} \Phi[s, v] - \sum_{v \in V} \Phi[v, s] = |\Phi|$$

(2) \Rightarrow (1): Nếu có một luồng f theo định nghĩa 2, ta xây dựng luồng dương Φ trên G như sau:

$$\Phi[u, v] := \max(f[u, v], 0)$$

Tức là ta chỉ giữ lại những giá trị $f[u, v]$ dương gán cho $\Phi[u, v]$. Ta chứng minh Φ thoả mãn hai điều kiện của luồng theo định nghĩa 1:

❖ Tính chất 1: $\forall u, v \in V$, rõ ràng $\Phi[u, v]$ không âm và $\Phi[u, v] = \max(f[u, v], 0) \leq c[u, v]$.

❖ Tính chất 2: $\forall u, v \in V$, ta có $f[u, v] = \Phi[u, v] - \Phi[v, u]$ bởi

Nếu $f[u, v] \geq 0$ thì $\Phi[u, v] = f[u, v]$ và $\Phi[v, u] = 0 \Rightarrow f[u, v] = \Phi[u, v] - \Phi[v, u]$

Nếu $f[u, v] < 0$ thì $\Phi[u, v] = 0$ và $\Phi[v, u] = f[v, u] = -f[u, v] \Rightarrow f[u, v] = \Phi[u, v] - \Phi[v, u]$

Ta lại có tổng luồng trên các cung đi vào $v = \sum_{u \in V} \Phi[u, v]$, tổng luồng trên các cung đi ra

khỏi $v = \sum_{u \in V} \Phi[v, u]$. Hiệu số giữa tổng luồng trên các cung đi vào v và tổng luồng trên các

cung đi ra khỏi $v = \sum_{u \in V} \Phi[u, v] - \sum_{u \in V} \Phi[v, u] = \sum_{u \in V} (\Phi[u, v] - \Phi[v, u]) = \sum_{u \in V} f[u, v] = 0$. Vậy

tổng luồng trên các cung đi ra khỏi v bằng tổng luồng trên các cung đi vào v

Về giá trị luồng, cũng từ đẳng thức $f[u, v] = \Phi[u, v] - \Phi[v, u]$, ta có

$$|\Phi| = \sum_{v \in V} \Phi[s, v] - \sum_{v \in V} \Phi[v, s] = \sum_{v \in V} (\Phi[s, v] - \Phi[v, s]) = \sum_{v \in V} f[s, v] = |f|$$

Định lý được chứng minh.

Định nghĩa 1 trực quan và dễ hiểu hơn định nghĩa 2, tuy nhiên định nghĩa 2 lại thích hợp hơn cho việc trình bày và chứng minh các thuật toán trong bài. Ta sẽ sử dụng định nghĩa 1 trong các hình vẽ và output (chỉ quan tâm tới các giá trị luồng dương) còn các khái niệm khi diễn giải thuật toán sẽ được hiểu theo định nghĩa 2.

10.1.5. Các tính chất cơ bản

Cho X và Y là hai tập con của đỉnh V , ta gọi khả năng thông qua từ X đến Y là:

$$c(X, Y) = \sum_{u \in X, v \in Y} c[u, v]$$

và giá trị luồng thông từ X sang Y là:

$$f(X, Y) = \sum_{u \in X, v \in Y} f[u, v]$$

Định lý 2:

❖ Với $\forall X \subseteq V$, ta có $f(X, X) = 0$

❖ Với $\forall X, Y \subseteq V$, ta có $f(X, Y) = -f(Y, X)$

❖ Với $\forall X, Y, Z \subseteq V$, $X \cap Y = \emptyset$, ta có $f(X, Z) + f(Y, Z) = f(X \cup Y, Z)$

❖ Với $\forall X \subseteq V \setminus \{s, t\}$, ta có $f(X, V) = 0$

Chứng minh:

❖ Với $\forall X \subseteq V$, ta có $f(X, X) = \sum_{u, v \in X} f[u, v]$, như vậy nếu $f[u, v]$ xuất hiện trong tổng thì $f[v, u]$ cũng xuất hiện, theo tính chất 2 của luồng, ta có $f(X, X) = 0$

❖ Với $\forall X, Y \subseteq V$, ta có $f(X, Y) = \sum_{u \in X, v \in Y} f[u, v]$, như vậy nếu $f[u, v]$ xuất hiện trong tổng thì $f[v, u]$ cũng xuất hiện, theo tính chất 2 của luồng, ta có $f(X, Y) = -f(Y, X)$

❖ Với $\forall X, Y \subseteq V$, ta có $f(X, Y) = \sum_{u \in X, v \in Y} f[u, v]$ và $f(Y, X) = \sum_{v \in Y, u \in X} f[v, u]$, nếu hạng tử $f[u, v]$ xuất hiện trong tổng thứ nhất thì hạng tử $f[v, u]$ xuất hiện trong tổng thứ hai và ngược lại.

Suy ra $f(X, Y) = -f(Y, X)$ theo tính chất 2 của luồng.

❖ $f(X \cup Y, Z) = \sum_{\substack{u \in X \cup Y \\ v \in Z}} f[u, v] = \sum_{\substack{u \in X \\ v \in Z}} f[u, v] + \sum_{\substack{u \in Y \\ v \in Z}} f[u, v] = f(X, Z) + f(Y, Z)$

❖ Với $\forall u \in V \setminus \{s, t\}$, ta có $f(\{u\}, V) = 0$ theo tính chất 3 của luồng, từ đó suy ra với $\forall X \subseteq V \setminus \{s, t\}$, ta có $f(X, V) = 0$. (Đi nhiên khi đó $f(V, X) = 0$)

Định lý 3: Giá trị luồng trên mạng bằng tổng luồng trên các cung đi vào đỉnh thu.

Chứng minh:

$$\begin{aligned} |f| &= f(\{s\}, V) && (\text{Định nghĩa}) \\ &= f(V, V) - f(V \setminus \{s\}, V) && (\text{Định lý 2}) \\ &= -f(V \setminus \{s\}, V) && (\text{Định lý 2}) \\ &= f(V, V \setminus \{s\}) && (\text{Định lý 2}) \\ &= f(V, \{t\}) + f(V, V \setminus \{s, t\}) && (\text{Định lý 2}) \\ &= f(V, \{t\}) && (\text{Định lý 2}) \end{aligned}$$

10.2. MẠNG THẶNG DƯ VÀ ĐƯỜNG TĂNG LUỒNG


10.2.1. Mạng thặng dư

Từ một mạng G và một luồng f trên G , ta xây dựng mạng $G_f = (V, E_f)$ với tập cạnh E_f được định nghĩa bằng tập cạnh E bỏ đi các cung bão hòa (một cung gọi là bão hòa nếu luồng trên cung đó đúng bằng khả năng thông qua):

$$E_f = \{(u, v) \mid (u, v \in V) \wedge (f[u, v] < c[u, v])\}$$

Khả năng thông qua của cung (u, v) trên G_f được tính bằng $c_f[u, v] = c[u, v] - f[u, v]$, s và t lần lượt được coi là điểm phát và điểm thu trên G_f .

Mạng G_f như vậy được gọi là mạng thặng dư (Residual Network) của mạng G sinh ra bởi luồng f . Hình 78 là ví dụ về một mạng thặng dư:



Hình 78: Mạng G và mạng thặng dư G_f tương ứng (ký hiệu $f[u,v]:c[u,v]$ chỉ luồng $f[u, v]$ và khả năng thông qua $c[u, v]$ trên cung (u, v))

Định lý 4: Cho mạng G và luồng f. Gọi f' là một luồng trên mạng G_f . Khi đó $(f+f')$ cũng là một luồng trên G với giá trị luồng bằng $|f|+|f'|$. Trong đó luồng $(f+f')$ được định nghĩa như sau:

$$(f+f')[u, v] = f[u, v] + f'[u, v] \quad (\forall u, v \in V)$$

Chứng minh: Ta chứng minh $(f+f')$ thoả mãn 3 tính chất của luồng.

- ❖ Với $\forall u, v \in V$. Vì $f[u, v] \leq c_f[u, v] = c[u, v] - f[u, v]$ nên $f[u, v] + f'[u, v] \leq c[u, v]$. Tính chất 1 được thoả mãn
- ❖ Với $\forall u, v \in V$. $(f+f')[u, v] = f[u, v] + f'[u, v] = -f[v, u] - f'[v, u] = -(f+f')[v, u]$. Tính chất 2 thoả mãn
- ❖ Với $\forall u \in V \setminus \{s, t\}$. $\sum_{v \in V} (f+f')[u, v] = \sum_{v \in V} (f[u, v] + f'[u, v]) = \sum_{v \in V} f[u, v] + \sum_{v \in V} f'[u, v] = 0$. Tính chất 3 được thoả mãn.

Về giá trị luồng. Ta có


$$|f+f'| = \sum_{v \in V} (f+f')[s, v] = \sum_{v \in V} (f[s, v] + f'[s, v]) = \sum_{v \in V} f[s, v] + \sum_{v \in V} f'[s, v] = |f| + |f'|$$

Định lý được chứng minh.

10.2.2. Đường tăng luồng

Cho mạng G và luồng f, một đường đi cơ bản từ A tới B trên mạng G_f gọi là một đường tăng luồng (Augmenting Path).

Với một đường tăng luồng P, ta đặt $\Delta_P := \min\{c_f(u, v) | (u, v) \in P\}$ là giá trị nhỏ nhất của các khả năng thông qua trên các cung trên P và gọi Δ_P là giá trị thặng dư (Residual capacity) của đường P.



Hình 79: Mạng thặng dư và đường tăng luồng

Định lý 5: Với P là một đường tăng luồng thì các giá trị $f_P[u, v]$ cho bởi :

$$f_P[u, v] = \begin{cases} \Delta_p, & \text{if } (u, v) \in P \\ -\Delta_p, & \text{if } (v, u) \in P \\ 0, & \text{otherwise} \end{cases}$$

là một luồng trên G_f .

Chứng minh: Để thấy f_P thoả mãn 3 tính chất của luồng.

Định lý 6: Cho mạng G và luồng f, nếu P là đường tăng luồng trên mạng thặng dư G_f thì phép đặt $f := (f + f_P)$ cho ta một luồng mới trên f có giá trị bằng $|f| + \Delta_p$.

Chứng minh: Dễ dàng suy ra từ định lý 5 và định lý 6.


Phép gán $f := (f + f_P)$ gọi là tăng luồng dọc theo đường P.

Ví dụ: Với mạng G, luồng f, và đồ thị thặng dư G_f như Hình 79, giả sử chọn đường đi $P = \langle 1, 3, 4, 2, 5, 6 \rangle$ làm đường tăng luồng với giá trị thặng dư $\Delta_p = 2$. Luồng f_P trên G_f sẽ là:

$$f_P[1, 3] = f_P[3, 4] = f_P[4, 2] = f_P[2, 5] = f_P[5, 6] = 2$$

$$f_P[3, 1] = f_P[4, 3] = f_P[2, 4] = f_P[5, 2] = f_P[6, 5] = -2$$

Cộng các giá trị luồng này vào giá trị luồng f tương ứng trên các cung, ta sẽ có luồng mới mang giá trị 9.



Hình 80: Luồng trên mạng G trước và sau khi tăng

10.3. THUẬT TOÁN FORD-FULKERSON (L.R.FORD & D.R.FULKERSON - 1962)

Thuật toán Ford-Fulkerson có thể viết:

```
<Khởi tạo một luồng f trên G, chẳng hạn luồng 0>
while <Tìm được đường tăng luồng P trên G_f> do f := (f + f_p);
<Output f>
```

Trước hết dễ thấy thuật toán Ford-Fulkerson trả về luồng tức là kết quả của thuật toán trả về thoả mãn 3 tính chất của luồng. Việc chứng minh luồng đó là cực đại đã xây dựng một định lý quan trọng về mối quan hệ giữa luồng cực đại và lát cắt hẹp nhất.

10.3.1. Lát cắt

Ta gọi một lát cắt (X, Y) là một cách phân hoạch tập đỉnh V thành 2 tập khác rỗng rời nhau X và Y . Lát cắt thoả mãn $s \in X$ và $t \in Y$ gọi là lát cắt $s-t$. Khả năng thông qua của lát cắt (X, Y) được định nghĩa bởi $c(X, Y) = \sum_{\substack{u \in X \\ v \in Y}} c[u, v]$. Luồng thông qua lát cắt (X, Y) định nghĩa bởi

$$f(X, Y) = \sum_{\substack{u \in X \\ v \in Y}} f[u, v].$$

Định lý 7: Cho mạng G và luồng f , khi đó luồng thông qua lát cắt $s-t$ bất kỳ bằng $|f|$.

Chứng minh: Với (X, Y) là một lát cắt $s-t$ bất kỳ,

$$\begin{aligned} f(X, Y) &= f(X, V) - f(X, V \setminus Y) && (\text{Định lý 2}) \\ &= f(X, V) - f(X, X) && (V \setminus Y = X) \\ &= f(X, V) && (\text{Định lý 2}) \\ &= f(s, V) + f(X \setminus \{s\}, V) && (\text{Định lý 2}) \\ &= f(s, V) && (\text{Định lý 2}, X \setminus \{s\} \text{ không chứa cả } s \text{ và } t) \\ &= |f| \end{aligned}$$

Định lý 8: Cho mạng G và luồng f , và (X, Y) là một lát cắt $s-t$, khi đó luồng thông qua lát cắt (X, Y) không vượt quá khả năng thông qua của lát cắt (X, Y): $f(X, Y) \leq c(X, Y)$.

Chứng minh: $f(X, Y) = \sum_{\substack{u \in X \\ y \in V}} f[u, v] \leq \sum_{\substack{u \in X \\ y \in V}} c[u, v] = c(X, Y)$

Hệ quả: Giá trị của một luồng f bất kỳ trên mạng G không vượt quá khả năng thông qua của một lát cắt $s-t$ bất kỳ.

Chứng minh: Suy ra trực tiếp từ định lý 8 và 9.

Định lý 9: (Định lý luồng cực đại và lát cắt $s-t$ hẹp nhất)

Nếu f là một luồng trên mạng $G = (V, E)$ với điểm phát s và điểm thu t , khi đó ba mệnh đề sau là tương đương:

- (1). f là luồng cực đại trên G

(2). Mạng thặng dư G_f không có đường tăng luồng

(3). Tồn tại (X, Y) là một lát cắt s-t để $|f| = c(X, Y)$

Chứng minh

(1) \Rightarrow (2)

Giả sử phản chứng rằng G_f có đường tăng luồng P thì ta có $(f + f_p)$ cũng là luồng trên G và giá trị luồng của $(f + f_p)$ lớn hơn f , trái với giả thiết f là luồng cực đại trên mạng

(2) \Rightarrow (3)

Nếu G_f không tồn tại đường tăng luồng thì ta đặt

❖ $X = \{u | \text{Tồn tại đường đi từ } s \text{ tới } u \text{ trên } G_f\}$

❖ $Y = V \setminus X$

Khi đó $X \cap Y = \emptyset$, $s \in X$, $t \notin X$ (do không có đường đi từ s tới t) nên $t \in Y$, nên (X, Y) là một lát cắt s-t. Với $\forall x \in X$ và $\forall y \in Y$, ta có $f[u, v] = c[u, v]$ bởi nếu $f[u, v] < c[u, v]$ thì có cung (u, v) trên G_f và như vậy từ s có thể đến được v , trái với cách xây dựng lát cắt. Từ đó suy ra $f(X, Y) = c(X, Y)$

(3) \Rightarrow (1)

Theo định lý 8, giá trị của mọi luồng trên $G \leq c(X, Y)$, nếu giá trị luồng f đúng bằng khả năng thông qua của lát cắt (X, Y) thì f phải là luồng cực đại trên mạng

Định nghĩa: Lát cắt s-t có khả năng thông qua nhỏ nhất (bằng giá trị luồng cực đại trên mạng) gọi là lát cắt s-t hẹp nhất của mạng G .


10.3.2. Cài đặt

Ta sẽ cài đặt thuật toán Ford-Fulkerson với Input và Output như sau

Input: file văn bản MAXFLOW.INP. Trong đó:

- ❖ Dòng 1: Chứa số đỉnh n (≤ 1000), số cạnh m của đồ thị, đỉnh phát s , đỉnh thu t theo đúng thứ tự cách nhau ít nhất một dấu cách
- ❖ m dòng tiếp theo, mỗi dòng có dạng ba số $u, v, c[u, v]$ cách nhau ít nhất một dấu cách thể hiện có cung (u, v) trong mạng và khả năng thông qua của cung đó là $c[u, v]$ ($c[u, v]$ là số nguyên dương không quá 10^6)

Output: file văn bản MAXFLOW.OUT, ghi luồng trên các cung và giá trị luồng cực đại tìm được



MAXFLOW.INP	MAXFLOW.OUT
6 8 1 6	f[1, 2] = 5
1 2 5	f[1, 3] = 4
1 3 5	f[2, 4] = 3
2 4 6	f[2, 5] = 2
2 5 3	f[3, 4] = 3
3 4 3	f[3, 5] = 1
3 5 1	f[4, 6] = 6
4 6 6	f[5, 6] = 3
5 6 6	Max Flow: 9

Ta sử dụng một số biến với các ý nghĩa như sau

- ❖ $c[1..n, 1..n]$: Ma trận biểu diễn khả năng thông qua của các cung trên mạng
- ❖ $f[1..n, 1..n]$: Ma trận biểu diễn luồng trên các cung
- ❖ Trace[1..n]: Dùng để lưu vết đường tăng luồng, thuật toán tìm đường tăng luồng sẽ sử dụng là thuật toán tìm kiếm theo chiều rộng (BFS)
- ❖ Ta có thể kiểm tra (u, v) có phải là cung trên mạng thặng dư G_f không bằng cách thức: $c[u, v] > f[u, v]$. Nếu (u, v) là cung trên G_f thì khả năng thông qua của nó là $c[u, v] - f[u, v]$.

P_4_10_1.PAS * Thuật toán Ford-Fulkerson

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program Max_Flow_by_Ford_Fulkerson;
const
  InputFile  = 'MAXFLOW.INP';
  OutputFile = 'MAXFLOW.OUT';
  max = 1000;
type
  TCapacities = array[1..max, 1..max] of Integer;
var
  c: TCapacities;
  f: TCapacities;
  Trace: array[1..max] of Integer;
  n, s, t: Integer;

procedure Enter; {Nhập dữ liệu}
var
  m, i, u, v: Integer;
  fi: Text;
begin
  Assign(fi, InputFile); Reset(fi);
  FillChar(c, SizeOf(c), 0);
  ReadLn(fi, n, m, s, t);
  for i := 1 to m do
    ReadLn(fi, u, v, c[u, v]);
  Close(fi);
end;

function FindPath: Boolean; {Tìm đường tăng luồng trên  $G_f$ , trả về True  $\Leftrightarrow$  tìm thấy}
var
  u, v: Integer;
  Queue: array[1..max] of Integer; {Hàng đợi dùng cho BFS}
  Front, Rear: Integer;
begin
  FillChar(Trace, SizeOf(Trace), 0);
  Front := 1; Rear := 1;
  Queue[1] := s;
  Trace[s] := n + 1; {Trace[v] = 0  $\Rightarrow$  v chưa thăm}
repeat

```

```

u := Queue[Front]; Inc(Front); {Lấy u khỏi Queue}
for v := 1 to n do
    if (Trace[v] = 0) and (c[u, v] > f[u, v]) then {Xét ∀v chưa thăm kề u trên G_f}
        begin
            Trace[v] := u;
            if v = t then {Đến được t thì thuật toán dừng}
                begin
                    FindPath := True; Exit;
                end;
            Inc(Rear); Queue[Rear] := v; {Đẩy v vào Queue}
        end;
    until Front > Rear;
    FindPath := False;
end;

procedure IncFlow; {Tăng luồng dọc đường tăng luồng: f := (f + f_p)}
var
    Delta, u, v: Integer;
begin
    {Tính Delta = Δ_p}
    Delta := MaxInt;
    v := t;
    repeat
        u := Trace[v];
        if c[u, v] - f[u, v] < Delta then Delta := c[u, v] - f[u, v];
        v := u;
    until v = s;
    {f := (f + f_p)}
    v := t;
    repeat
        u := Trace[v];
        f[u, v] := f[u, v] + Delta;
        f[v, u] := f[v, u] - Delta;
        v := u;
    until v = s;
end;

procedure PrintResult;
var
    u, v: Integer;
    m: Integer;
    fo: Text;
begin
    Assign(fo, OutputFile); Rewrite(fo);
    m := 0;
    for u := 1 to n do
        for v := 1 to n do
            if f[u, v] > 0 then {Chi quan tâm đến những cung có luồng dương}
                begin
                    WriteLn(fo, 'f['', u, ', ', v, ']' = ', f[u, v]);
                    if u = s then m := m + f[s, v];
                end;
    WriteLn(fo, 'Max Flow: ', m);
    Close(fo);
end;

begin
    Enter;
    FillChar(f, SizeOf(f), 0);
    repeat
        if not FindPath then Break;
        IncFlow;
    until False;

```

```

PrintResult;
end.

```

Định lý 10: (Tính nguyên): Nếu tất cả các khả năng thông qua là số nguyên thì thuật toán trên luôn tìm được luồng cực đại với luồng trên cung là các số nguyên.

Chứng minh: Ban đầu khởi tạo luồng 0 thì tức các luồng trên cung là nguyên. Mỗi lần tăng luồng lên một lượng bằng trọng số nhỏ nhất trên các cung của đường tăng luồng cũng là số nguyên nên cuối cùng luồng cực đại sẽ phải có luồng trên các cung là nguyên.

Định lý 11: (Độ phức tạp tính toán): Edmonds và Karp chứng minh rằng nếu dùng thuật toán BFS để tìm đường tăng luồng trên mạng được biểu diễn theo kiểu danh sách kề thì có thể cài đặt thuật toán Ford-Fulkerson bằng giải thuật có độ phức tạp $O(nm^2)$. Tuy nhiên nếu khả năng thông qua trên các cung của mạng là số nguyên thì có một cách đánh giá khác dựa trên giá trị luồng cực đại: Độ phức tạp tính toán của thuật toán Ford-Fulkerson là $O(|f^*| \cdot m)$ với $|f^*|$ là giá trị luồng cực đại trên mạng.

10.4. THUẬT TOÁN PREFLOW-PUSH (GOLDBERG - 1986)

Thuật toán Ford-Fulkerson không những là một cách tiếp cận thông minh mà việc chứng minh tính đúng đắn của nó cho ta nhiều kết quả thú vị về mối liên hệ giữa luồng cực đại và lát cắt hẹp nhất. Tuy vậy độ phức tạp tính toán của thuật toán Ford-Fulkerson là khá lớn, dẫn tới những khó khăn khi thực hiện với dữ liệu lớn. Trong phần này ta sẽ trình bày một lớp các thuật toán nhanh nhất cho tới nay để giải bài toán luồng cực đại, tên chung của các thuật toán này là Preflow-push.

Ta có thể hình dung mạng như một hệ thống đường ống dẫn nước từ với điểm phát s tới điểm thu t, các cung là các đường ống, khả năng thông qua là lưu lượng đường ống có thể tải. Nước chảy theo nguyên tắc từ chỗ cao về chỗ thấp. Với một lượng nước lớn phát ra từ s tới một đỉnh v, nếu có cách chuyển lượng nước đó sang địa điểm khác thì không có vấn đề gì, nếu không thì có hiện tượng “quá tải” xảy ra tại v, ta “dâng cao” điểm v để lượng nước đó đổ sang điểm khác (có thể đó ngược về s). Cứ tiếp tục quá trình như vậy cho tới khi không còn hiện tượng quá tải ở bất cứ điểm nào. Cách tiếp cận này hoàn toàn khác với thuật toán Ford-Fulkerson: thuật toán Ford-Fulkerson cố gắng tìm một dòng chảy phụ (f_p) từ s tới t và thêm dòng chảy này vào luồng hiện có đến khi không còn dòng chảy phụ nữa.

10.4.1. Preflow

Cho một mạng $G = (V, E)$. Ta gọi một preflow là một phép gán cho mỗi cung $(u, v) \in E$ một số thực $f[u, v]$ thoả mãn 3 tính chất:

- ❖ Tính chất 1 (Capacity constraint): preflow trên mỗi cung cũng không được vượt quá khả năng thông qua của cung đó: $f[u, v] \leq c[u, v], (\forall u, v \in V)$
- ❖ Tính chất 2 (Skew symmetry): preflow trên cung (u, v) và preflow trên cung (v, u) có cùng giá trị tuyêt đối nhưng trái dấu nhau: $f[u, v] = -f[v, u]$

❖ Tính chất 3: Tổng luồng trên các cung đi vào mỗi đỉnh $v \in V \setminus \{s\}$ là số không âm:

$$\sum_{u \in V} f[u, v] \geq 0$$

Ta dùng biến $\text{FlowIn}[v]$ để lưu tổng luồng trên các cung đi vào đỉnh v : $\text{FlowIn}[v] = \sum_{u \in V} f[u, v]$.

Đỉnh $v \in V \setminus \{s, t\}$ được gọi là quá tải (overflow) nếu $\text{FlowIn}[v] > 0$. Lưu ý rằng khái niệm quá tải chỉ đề cập tới các đỉnh không phải đỉnh phát cũng không phải đỉnh thu.

Định nghĩa về preflow tương tự như định nghĩa luồng, chỉ khác nhau ở tính chất 3. Vì vậy ta cũng có khái niệm mạng thặng dư G_f ứng với preflow tương tự như đối với luồng.

10.4.2. Thao tác khởi tạo (Init)

Ta gọi phép cho tương ứng mỗi đỉnh v với một số tự nhiên $h[v]$ là một hàm độ cao. Hàm độ cao được khởi tạo bằng 0 trên các đỉnh không phải đỉnh phát và bằng n tại đỉnh phát:

```
for ( $\forall v \in V$ ) do
  if  $v = s$  then  $h[v] := n$ 
  else  $h[v] := 0$ 
```

Sau thao tác khởi tạo hàm độ cao, ta khởi tạo preflow:

```
for ( $\forall u, v \in V, u \neq s$  và  $v \neq s$ ) do  $f[u, v] := 0$ 
for ( $\forall u \in V, u \neq s$ ) do
begin
   $f[s, u] := c[s, u];$ 
   $f[u, s] := -c[s, u];$ 
end;
```

Cuối cùng, ta phải tính các giá trị $\text{FlowIn}[.]$ tương ứng với preflow vừa khởi tạo

```
for ( $\forall u \in V$ ) do  $\text{FlowIn}[u] := f[s, u]$ 
```

10.4.3. Thao tác Push

Thao tác Push(u, v) được áp dụng khi 3 điều kiện sau được thoả mãn:

- ❖ u bị quá tải: $(u \neq s)$ and $(u \neq t)$ and $(\text{FlowIn}[u] > 0)$
- ❖ (u, v) là cung trên G_f : $c_f[u, v] = c[u, v] - f[u, v] > 0$
- ❖ Đỉnh u cao hơn đỉnh v : $h[u] > h[v]$

Khi đó thao tác Push(u, v) sẽ thực hiện qua các bước sau:

- ❖ Đặt $\Delta := \min(\text{FlowIn}[u], c_f[u, v])$
- ❖ Đặt $f[u, v] := f[u, v] + \Delta$
- ❖ Đặt $f[v, u] := f[v, u] - \Delta$
- ❖ $\text{FlowIn}[u] := \text{FlowIn}[u] - \Delta$
- ❖ $\text{FlowIn}[v] := \text{FlowIn}[v] + \Delta$

Bản chất của thao tác Push(u, v) là chuyển một lượng nước Δ đọng tại u (do quá tải) sang đỉnh v . Dễ thấy cả ba tính chất của Preflow vẫn được duy trì sau thao tác Push.

10.4.4. Thao tác Lift

Thao tác Lift(u) được áp dụng khi hai điều kiện sau được thoả mãn:

- ❖ u bị quá tải: $(u \neq s)$ and $(u \neq t)$ and $(\text{FlowIn}[u] > 0)$
- ❖ u không chuyển được luồng xuống nơi thấp hơn: $(u, v) \in E_f \Rightarrow h[u] \leq h[v]$

Khi đó thao tác Lift(u) đặt $h[u] := 1 + \min\{h[v] | (u, v) \in E_f\}$

Bản chất của thao tác Lift là khi đỉnh u bị quá tải và cũng không chuyển tải được cho đỉnh khác thì ta đẩy đỉnh u lên cao hơn đỉnh thấp nhất có thể chuyển tải sang đúng một đơn vị độ cao.

10.4.5. Thuật toán Preflow-push

Thuật toán Preflow-push có thể viết:

```
Init;
while <Còn đỉnh bị quá tải u> do
    if < $\exists v \in V$  để có thể thực hiện thao tác Push( $u, v$ )> then
        Push( $u, v$ )
    else
        Lift( $u$ );
<Output f>
```

Việc chứng minh tính đúng đắn của thuật toán Preflow-push có thể suy ra từ các định lý sau:

Định lý 12: Độ cao của các đỉnh không bao giờ bị giảm đi cả.

Chứng minh: Dễ thấy bởi chỉ có thao tác Lift thay đổi độ cao của đỉnh, thao tác này chỉ nâng độ cao một đỉnh lên chứ không giảm đi.

Định lý 13: Các độ cao $h[.]$ luôn thoả mãn ràng buộc: $\forall (u, v) \in E_f$ thì $h[u] \leq h[v] + 1$. Ràng buộc này gọi là ràng buộc độ cao

Chứng minh: Rõ ràng tại bước khởi tạo, các độ cao $h[.]$ thoả mãn:

$$\forall (u, v) \in E_f \Rightarrow h[u] \leq h[v] + 1.$$

Thao tác Lift(u)

- ❖ Với mọi cung $(u, v) \in E_f$, thao tác Lift đặt $h[u] := 1 + \min\{h[v] | (u, v) \in E_f\}$ đẩy u lên cao hơn nhưng không quá $h[v] + 1$.
- ❖ Với mọi cung $(w, u) \in E_f$, trước khi vào thủ tục Lift thì $h[w] \leq h[u] + 1$ theo giả thiết, vậy nếu đẩy u lên cao hơn thì $h[w]$ vẫn không quá $h[u] + 1$.

Thao tác Push(u, v)

Chú ý rằng thao tác Push(u, v) chỉ thực hiện được nếu $h[u] > h[v]$. Thao tác Push trước hết thêm vào E_f cung (v, u) khi đó ta có $h[v] < h[u]$, ràng buộc độ cao được duy trì. Sau đó thao tác Push có thể dỡ bỏ cung (u, v) từ E_f , việc dỡ bỏ này cũng bỏ luôn ràng buộc độ cao ($h[u]$ phải $\leq h[v] + 1$).

Định lý được chứng minh.

Nói thêm rằng có thể chứng minh được thao tác Push(u, v) chỉ thực hiện được nếu u cao hơn v đúng 1 đơn vị ($h[u] = h[v] + 1$). Điều này có thể suy ra từ ràng buộc độ cao $h[u] \leq h[v] + 1$ và điều kiện $h[u] > h[v]$ của thao tác Push.

Định lý 14: Cho $G = (V, E)$ là một mạng với điểm phát s và điểm thu t , gọi f là một preflow trên G . Khi đó nếu tồn tại một hàm độ cao h trên V thoả mãn ràng buộc độ cao thì không có đường đi từ s tới t trong G_f .

Chứng minh: Giả sử phản chứng rằng có đường đi $P = \langle s = v_0, v_1, \dots, v_k = t \rangle$, không giảm tính tổng quát, có thể coi P là đường đi cơ bản, tức là $k < n$. Từ ràng buộc độ cao ta có:

$$h[s] = h[v_0] \leq h[v_1] + 1 \leq h[v_2] + 2 \leq \dots \leq h[v_k] + k = h[t] + k.$$

Bởi $h[t] = 0$, ta có $h[s] \leq k < n$. Mâu thuẫn với việc $h[s]$ được khởi tạo bằng n và không bao giờ bị thay đổi cả.

Định lý 15: Thuật toán Preflow-push là đúng đắn và có độ phức tạp tính toán $O(n^2m)$.

Chứng minh: Có thể chứng minh thuật toán Preflow-Push dừng trong thời gian $O(n^2m)$. Nếu thuật toán dừng thì không còn đỉnh nào bị quá tải, ta có f là luồng. Theo định lý 14, không còn đường đi từ s tới t trên G_f . Áp dụng định lý 9, ta có f là luồng cực đại trên mạng.

Có thể đưa vào một số cải tiến, chẳng hạn tổ chức danh sách các đỉnh quá tải theo dạng Queue, sau đó với mỗi thao tác, thuật toán đẩy thêm những đỉnh quá tải mới phát hiện trong mỗi thao tác Push vào chờ trong Queue để không tốn thời gian tìm đỉnh quá tải. Thuật toán tiến hành theo cách này có độ phức tạp tính toán $O(n^3)$.

10.4.6. Cài đặt

Chương trình cài đặt thuật toán Preflow-push có Input và Output giống chương trình cài đặt thuật toán Ford-Fulkerson.

Các đỉnh quá tải sẽ được lưu trong một cấu trúc dữ liệu Queue dạng danh sách vòng (xem PHẦN 2, §5, 5.2.2). Mảng Boolean InQueue[1..n], trong đó InQueue[u] cho biết đỉnh u đã có mặt trong Queue chưa. Hàm Discharge(u) được thực hiện với một đỉnh u bị quá tải, hàm này sẽ cố gắng chuyển tải từ u sang những đỉnh v kề với u trên G_f qua thao tác Push. Nếu giảm tải được cho u thì hàm trả về True, ngược lại hàm trả về False để báo cho chương trình biết phải thực hiện thao tác Lift(u).

```

P_4_10_2.PAS * Thuật toán Preflow-push
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program MaximumFlowByPreFlowPush;
const
  InputFile = 'MAXFLOW.INP';
  OutputFile = 'MAXFLOW.OUT';
  max = 1000;
type
  TCapacities = array[1..max, 1..max] of Integer;
var
  c: TCapacities;
  f: TCapacities;
  FlowIn: array[1..max] of Integer;
  h: array[1..max] of Integer;
  Queue: array[0..max - 1] of Integer;
  InQueue: array[1..max] of Boolean;
  n, s, t, Front, Rear: Integer;

procedure Enter; {Nhập dữ liệu}
var

```

```

m, i, u, v: Integer;
fi: Text;
begin
  Assign(fi, InputFile); Reset(fi);
  FillChar(c, SizeOf(c), 0);
  ReadLn(fi, n, m, s, t);
  for i := 1 to m do
    ReadLn(fi, u, v, c[u, v]);
  Close(fi);
end;

function OverFlow(u: Integer): Boolean; {OverFlow[u] ⇔ u bị quá tải}
begin
  OverFlow := (u < s) and (u > t) and (FlowIn[u] > 0)
end;

procedure Init; {Khởi tạo}
var
  v: Integer;
begin
  FillChar(f, SizeOf(f), 0);
  FillChar(InQueue, SizeOf(InQueue), False);
  FillChar(h, SizeOf(h), 0); {Khởi tạo hàm độ cao}
  h[s] := n;
  Rear := n - 1;
  for v := 1 to n do {Cho s phát hết công suất lên các cung liên thuộc (s, v)}
    begin
      f[s, v] := c[s, v];
      f[v, s] := -c[s, v];
      FlowIn[v] := c[s, v];
      if OverFlow(v) then {Nếu v bị quá tải thì đưa v vào Queue}
        begin
          Rear := (Rear + 1) mod n;
          Queue[Rear] := v;
          InQueue[v] := True;
        end;
    end;
  Front := 0;
end;

procedure PushToQueue(u: Integer); {Đẩy một đỉnh quá tải u vào Queue}
begin
  if not InQueue[u] then
    begin
      Rear := (Rear + 1) mod n;
      Queue[Rear] := u;
      InQueue[u] := True;
    end;
end;

function PopFromQueue: Integer; {Lấy một đỉnh quá tải ra khỏi Queue, trả về trong kết quả hàm}
var
  u: Integer;
begin
  u := Queue[Front];
  Front := (Front + 1) mod n;
  InQueue[u] := False;
  PopFromQueue := u;
end;

function Discharge(u: Integer): Boolean; {Hàm Discharge có gắng giảm tải cho đỉnh quá tải u}
var
  v: Integer;

```

```

Delta: Integer;
Pushed: Boolean;
begin
  Pushed := False;
  for v := 1 to n do
    if (c[u, v] > f[u, v]) and (h[u] > h[v]) then {Điều kiện để thực hiện Push(u, v)}
      begin
        {Thực hiện thao tác Push(u, v)}
        Delta := c[u, v] - f[u, v];
        if FlowIn[u] < Delta then Delta := FlowIn[u];
        f[u, v] := f[u, v] + Delta;
        f[v, u] := f[v, u] - Delta;
        FlowIn[u] := FlowIn[u] - Delta;
        FlowIn[v] := FlowIn[v] + Delta;
        if OverFlow(v) then PushToQueue(v); {Thao tác Push(u, v) có thể sinh ra đỉnh quá tải mới v}
        Pushed := True; {Đặt cờ báo u đã được giảm tải}
      end;
  Discharge := Pushed;
end;

function Lift(u: Integer): Boolean; {Thao tác Lift}
var
  v, MinH: Integer;
begin
  MinH := MaxInt;
  for v := 1 to n do
    if (c[u, v] > f[u, v]) and (h[v] < MinH) then
      MinH := h[v];
  h[u] := MinH + 1;
end;

procedure Preflowpush; {Thuật toán Preflow-push}
var
  u: Integer;
begin
  while Front <> (Rear + 1) mod n do
    begin
      u := PopFromQueue;
      if not Discharge(u) then Lift(u);
      if FlowIn[u] > 0 then PushToQueue(u); {Nếu Discharge(u) không chuyển tải được hết cho u thì u vẫn quá tải}
    end;
end;

procedure PrintResult; {In kết quả}
var
  u, v: Integer;
  m: Integer;
  fo: Text;
begin
  Assign(fo, OutputFile); Rewrite(fo);
  m := 0;
  for u := 1 to n do
    for v := 1 to n do
      if f[u, v] > 0 then
        begin
          WriteLn(fo, 'f[', u, ', ', v, '] = ', f[u, v]);
          if u = s then m := m + f[s, v];
        end;
  WriteLn(fo, 'Max Flow: ', m);
  Close(fo);
end;

begin

```

```

Enter;
Init;
Preflowpush;
PrintResult;
end.

```


Thực ra tên gọi preflow-push là một tên gọi chung cho một lớp thuật toán mà ý tưởng ban đầu của nó được phát triển bởi Karzanov năm 1974, mô hình chung nhất cho thuật toán preflow-push đầu tiên là generic preflow-push được Goldberg đưa ra năm 1987 trong luận văn PhD của ông. Sau đó có rất nhiều cải tiến từ mô hình này chẳng hạn: FIFO Preflow-push sử dụng hàng đợi như chúng ta cài đặt ở trên, Highest Label Preflow-push (luôn chọn đỉnh quá tải nằm ở độ cao nhất để giảm tải trước), Lift-to-front Preflow-push, ... Các bạn có thể tham khảo trong các tài liệu chuyên sâu về bài toán luồng.

10.5. MỘT SỐ MÔ RỘNG

Sau khi đã hiểu cẩn kẽ hai định nghĩa về luồng dương và luồng (mục 10.1), dưới đây ta sẽ sử dụng định nghĩa 1 về luồng cho dễ trình bày các khái niệm.

10.5.1. Mạng với nhiều điểm phát và nhiều điểm thu

Xét mạng G với p điểm phát $s[1], s[2], \dots, s[p]$ và q điểm thu $t[1], t[2], \dots, t[q]$. Một luồng có thể đi từ một điểm phát bất kỳ đến một điểm thu bất kỳ, được định nghĩa tương tự như trong bài toán luồng cực đại. Bài toán tìm luồng cực đại từ các điểm phát đến các điểm thu có thể đưa về bài toán với một điểm phát và một điểm thu bằng cách đưa vào thêm một điểm phát giả s và một điểm thu giả t . Đỉnh s được nối tới tất cả các điểm phát $s[1], \dots, s[p]$ và đỉnh t được nối từ tất cả các đỉnh thu $t[1], \dots, t[q]$ bằng cung có khả năng thông qua là $+\infty$ (Hình 81), sau đó tìm luồng cực đại trên mạng và cuối cùng dỡ bỏ hai đỉnh giả cũng như các cung giả mới thêm vào.




Hình 81: Mạng giả của mạng có nhiều điểm phát và nhiều điểm thu

10.5.2. Mạng với khả năng thông qua của các đỉnh và các cung

Giả sử trong mạng G , ngoài khả năng thông qua của các cung $c[u, v]$, mỗi đỉnh còn được gán một số không âm $d[.]$ là khả năng thông qua của đỉnh đó. Luồng được định nghĩa tương tự nhưng thêm ràng buộc: tổng luồng đi vào mỗi đỉnh v không được vượt quá $d[v]$:

$$\sum_{u \in V} f[u, v] \leq d[v].$$

Nếu ta thay mỗi đỉnh v trên đồ thị bằng hai đỉnh v_{in} , v_{out} và một cung (v_{in}, v_{out}) có khả năng thông qua là $d[v]$, sau đó thay mỗi cung đi vào v thành cung đi vào v_{in} và thay mỗi cung đi ra từ v bằng cung đi ra từ v_{out} (Hình 82) thì bài toán tìm luồng cực đại trên mạng G có thể giải quyết bằng cách tìm luồng cực đại trên mạng giả này và gán luồng trên mỗi cung (u, v) trên mạng ban đầu bằng giá trị luồng trên cung (u_{out}, v_{in}) trên mạng giả.




Hình 82: Thay một đỉnh u bằng hai đỉnh u_{in} , u_{out}

10.5.3. Mạng với ràng buộc luồng thông qua của các cung bị chặn cả trên và dưới

Xét mạng G trong đó mỗi cung (u, v) ngoài khả năng thông qua $c[u, v]$ còn được gán một số không âm $d[u, v]$ là cận dưới của luồng. Bài toán đặt ra là có tồn tại luồng tương thích trên G hay không. (Luồng tương thích trên G là một phép gán cho mỗi cung (u, v) một giá trị $f[u, v]$ thỏa mãn: $d[u, v] \leq f[u, v] \leq c[u, v]$).

Đưa vào mạng G một đỉnh phát giả s' và một đỉnh thu giả t' , xây dựng một mạng giả G' theo quy tắc: với mỗi cung (u, v) trên G sẽ tương ứng với 3 cung trên G' : cung (s', v) và (u, t') với khả năng thông qua là $d[u, v]$, cung (u, v) với khả năng thông qua là $c[u, v] - d[u, v]$. Ngoài ra thêm vào một cung (t, s) trên G' với khả năng thông qua $+\infty$ (Hình 83).



Hình 83: Mạng giả của mạng có khả năng thông qua của các cung bị chặn hai phía

Định lý 16: Điều kiện cần và đủ để tồn tại luồng tương thích trên G là trên G' phải tồn tại luồng từ s' tới t' với giá trị luồng là $\sum_{(u,v) \in E} d[u, v]$.

Dễ thấy nếu tồn tại luồng trên G' từ s' tới t' với giá trị luồng $\sum_{(u,v) \in E} d[u, v]$ thì luồng đó phải là luồng cực đại trên G' . Để chỉ ra luồng tương thích trên G , ta chỉ việc lấy giá trị luồng trên cung (u, v) của G' cộng thêm $d[u, v]$ là được luồng trên cung (u, v) của G .

Bài tập:**Bài 1**

Cho một đồ thị gồm n đỉnh và m cạnh và 2 đỉnh A, B . Hãy tìm cách bỏ đi một số ít nhất các cạnh để không còn đường đi từ A tới B .

Hướng dẫn: Coi $G = (V, E)$ là mạng với điểm phát A và điểm thu B , loại bỏ tất cả các cung đi vào A và các cung đi ra khỏi B , đặt khả năng thông qua của các cung đều bằng 1, tìm luồng cực đại trên mạng và lát cắt s-t hẹp nhất (X, Y), những cạnh nối giữa X và Y là những cạnh cần bỏ.

Bài 2: Hệ đại diện phân biệt

Một lớp học có n bạn nam, n bạn nữ. Cho m món quà lưu niệm, ($n \leq m$). Mỗi bạn có sở thích về một số món quà nào đó. Hãy tìm cách phân công mỗi bạn nam tặng một món quà cho một bạn nữ thỏa mãn:

- ❖ Mỗi bạn nam chỉ tặng quà cho đúng một bạn nữ
- ❖ Mỗi bạn nữ chỉ nhận quà của đúng một bạn nam
- ❖ Bạn nam nào cũng đi tặng quà và bạn nữ nào cũng được nhận quà, món quà đó phải hợp sở thích của cả hai người.
- ❖ Món quà nào đã được một bạn nam chọn thì bạn nam khác không được chọn nữa.

Hướng dẫn: Xây dựng mạng $G = (V, E)$, trong đó V gồm 4 lớp đỉnh:

- ❖ Lớp đỉnh $A[1..n]$, mỗi đỉnh tượng trưng cho một bạn nam
- ❖ Lớp đỉnh $B[1..m]$, mỗi đỉnh tượng trưng cho một món quà
- ❖ Lớp đỉnh $C[1..m]$, mỗi đỉnh tượng trưng cho một món quà
- ❖ Lớp đỉnh $D[1..n]$, mỗi đỉnh tượng trưng cho một bạn nữ.

Tập các cung E được xây dựng như sau:

- ❖ Một cung nối từ lớp A tới lớp B tương ứng với một bạn nam và một món quà hợp sở thích của bạn nam đó.
- ❖ Một cung nối từ lớp B sang lớp C nối một đỉnh tượng trưng cho một món quà từ lớp B tới đỉnh tượng trưng cho chính món quà đó ở lớp C .
- ❖ Một cung nối từ lớp C sang lớp D tương ứng với một món quà và một bạn nữ thích món quà đó.
- ❖ Tất cả các cung đều có khả năng thông qua bằng 1

Tìm luồng cực đại trên mạng, nếu giá trị luồng bằng m thì cách phân công là tồn tại và có thể chỉ ra bằng cách: Trên những cung có luồng đi qua (giá trị luồng đi qua bằng 1 theo định lý về tính nguyên), mỗi cung nối từ lớp A sang lớp B tương đương với một bạn nam và một món quà bạn nam đó sẽ chọn, mỗi cung từ lớp C sang lớp D tương đương với một món quà và một bạn nữ sẽ nhận món quà đó.

Bài 3: Minimum Path Cover

Cho một đồ thị $G = (V, E)$ có hướng và không có chu trình (Directed Acyclic Graph-DAG), một phủ đường (Path Cover) là một tập P gồm các đường đi trên G thỏa mãn: Với mọi đỉnh $v \in V$, tồn tại duy nhất một đường đi $p \in P$ chứa v . Đường đi có thể bắt đầu và kết thúc ở bất cứ đâu, có tính cả đường đi độ dài 0 (chỉ gồm 1 đỉnh).

Hãy tìm phủ đường gồm ít đường đi nhất ($|P| \rightarrow \min$).


Hướng dẫn:

Giả sử $V = \{1, \dots, n\}$. Xây dựng đồ thị $G' = (V', E')$, trong đó:

$$V' = \{x_0, x_1, \dots, x_n\} \cup \{y_0, y_1, \dots, y_n\}$$

$$E' = \{(x_0, x_i) | i \in V\} \cup \{(y_j, y_0) | j \in V\} \cup \{(x_i, y_j) | (i, j) \in E\}$$

Tìm luồng nguyên cực đại trên mạng G' với đỉnh phát x_0 , đỉnh thu y_0 và tất cả khả năng thông qua của các cung được gán bằng 1. Mỗi cung (x_i, y_j) trên G' có luồng đi qua sẽ tương ứng với một cung (i, j) của G được chọn vào một đường đi nào đó trong P . Cuối cùng, những đỉnh nào của G chưa thuộc P sẽ được đưa nốt vào P với tư cách là đường đi độ dài 0.



Trước hết ta có nhận xét rằng có một sự tương ứng giữa phủ đường trên G và luồng nguyên cực đại trên G' . Thật vậy:

- ❖ Với mỗi đường đi $\langle v[1], v[2], \dots, v[q] \rangle$ của P , với mỗi đỉnh $v[j]$ ($j=1, \dots, q$), ta có thể đầy 1 đơn vị luồng từ x_0 tới $x_{v[j]}$ qua $y_{v[j+1]}$ đến y_0 . Để thấy luồng trên G' được xây dựng như vậy là thỏa mãn định nghĩa luồng, bởi các đường đi trong P đôi một không có đỉnh chung.
- ❖ Ngược lại, với một luồng nguyên trên G' , ta bỏ qua không xét những cung nối từ x_0 và những cung nối tới y_0 trên G' , theo định lý về tính nguyên, luồng trên mỗi cung của G' hoặc bao hoà (luồng = khả năng thông qua = 1), hoặc bằng 0. Khởi tạo P gồm n đỉnh $\in V$ và chưa có cung nào, có thể coi P gồm n đường đi độ dài 0. Mỗi khi đưa vào P một cung (u, v) tương ứng với một cung bao hoà (x_u, y_v) trên G' , cung đó sẽ nối 2 đường trên P thành một đường. Tại sao vậy? bởi trên G' , mỗi đỉnh x_u có nhiều nhất một cung bao hoà đi ra và mỗi đỉnh y_v có nhiều nhất một cung bao hoà đi vào, vậy nên nếu thêm vào P cung (u, v) tương ứng với cung bao hoà (x_u, y_v) trên G' sẽ không bao giờ làm cho bán bắc ra của u cũng như bán bắc vào của v vượt quá 1 (tức là không tạo thành ngã rẽ). Từ đó suy ra, theo

thuật toán, mỗi lần thêm vào P một cung thì số đường đi trong P sẽ giảm đi 1. Vì số cung bao hoà dạng (x_u, y_v) trên G' đúng bằng giá trị luồng cực đại trên mạng G' nên số đường đi trên P xây dựng theo cách trên sẽ là $n - f$ với f là giá trị luồng thông qua mạng G'. Việc cực tiểu hoá số đường đi trong P tương đương với cực đại hoá giá trị luồng qua mạng G', tức là phủ đường suy ra từ luồng cực đại là phủ đường tối thiểu.

Thuật toán trên không thực hiện được trong trường hợp đồ thị có chu trình, hãy tự giải thích tại sao. Cho tới nay, người ta vẫn cho rằng bài toán tìm phủ đường cực tiểu trong trường hợp đồ thị tổng quát là NP-Hard, có nghĩa là một thuật toán với độ phức tạp đa thức để giải quyết bài toán phủ đường cực tiểu trên đồ thị tổng quát sẽ là một phát minh lớn và đáng ngạc nhiên.

Bài 4: The minimum cut

Ta quan tâm đến đồ thị vô hướng liên thông $G = (V, E)$, các cạnh được gán trọng số không âm, bài toán đặt ra là hãy phân hoạch tập đỉnh V thành hai tập khác rỗng rời nhau X và Y sao cho tổng trọng số các cạnh nối giữa X và Y là nhỏ nhất có thể. Cách phân hoạch này gọi là lát cắt tổng quát hẹp nhất của G. Ký hiệu $\text{MinCut}(G)$.

Hướng dẫn:

Gọi c là ma trận trọng số của G , $c[u, v] = 0$ nếu (u, v) không là cạnh.

Bài toán có thể phát biểu: Tìm $X \subseteq V$ và $Y \subseteq V$ thoả mãn:

$$(X \cup Y = V) \text{ and } (X \cap Y = \emptyset) \text{ and } (c(X, Y) \rightarrow \min)$$

Cách 1: Một cách tệ nhất có thể thực hiện là thử tất cả các cặp đỉnh (s, t) . Với mỗi lần thử ta cho s làm điểm phát và t làm điểm thu trên mạng G , tìm luồng cực đại và lát cắt $s-t$ hẹp nhất. Cuối cùng là chọn lát cắt $s-t$ có trọng số nhỏ nhất trong tất cả các lần thử. Cách này có thể nói là rất chậm (mất $n.(n-1)/2$ lần tìm luồng cực đại) nên không khả thi với dữ liệu lớn. Dưới đây ta trình bày cách làm khác.

Với hai điểm u và v của G , ta gọi đồ thị $G/\{u, v\}$ là đồ thị tạo thành từ G bằng cách chia 2 đỉnh u và v thành 1 đỉnh uv . Trọng số của các cạnh (uv, w) được tính bằng tổng $c[u, w] + c[v, w]$. Khi đó ta có định lý:

Định lý 17: Với s và t là hai đỉnh bất kỳ của G , khi đó $\text{MinCut}(G)$ có thể thu được bằng cách lấy lát cắt nhỏ hơn trong hai lát cắt:

❖ Lát cắt $s-t$ hẹp nhất: Coi s và t lần lượt là điểm phát và điểm thu trên G , lát cắt $s-t$ hẹp nhất có thể xác định bằng việc giải quyết bài toán luồng cực đại trên G với c là ma trận khả năng thông qua.

❖ Lát cắt tổng quát hẹp nhất trên $G/\{s, t\}$: $\text{MinCut}(G/\{s, t\})$.

Chứng minh: Lát cắt tổng quát hẹp nhất trên G có thể đưa s và t vào hai thành phần liên thông khác nhau hoặc đưa chúng vào cùng một thành phần liên thông. Trong trường hợp 1, $\text{MinCut}(G)$ là lát cắt $s-t$ hẹp nhất, trường hợp 2, $\text{MinCut}(G)$ là $\text{MinCut}(G/\{s, t\})$.

Đến đây ta có một thuật toán tốt hơn:

Cách 2: Nếu đồ thị có đúng 2 đỉnh thì chỉ việc cắt rời hai đỉnh, nếu không thì chọn lấy hai đỉnh phân biệt s, t bất kỳ lần lượt là điểm phát và điểm thu, tìm luồng cực đại trên mạng và ghi nhận lại lát cắt s-t. Sau đó chập s và t lại thành một đỉnh st và lặp lại. Cuối cùng là chỉ ra lát cắt tổng quát hẹp nhất trong số các lát cắt đã ghi nhận. Cách 2 tốt hơn cách 1 ở chỗ: thay vì $n(n-1)/2$ lần tìm luồng cực đại ta chỉ cần $n-1$ lần tìm luồng cực đại. Tuy vậy cách này vẫn chưa phải thật tốt.

Có thể nhận xét rằng tại mỗi bước của cách giải 2, ta có thể chọn hai đỉnh s và t bất kỳ, miễn sao $s \neq t$. Vậy ta sẽ tìm một “chiến thuật” chọn hai đỉnh s và t một cách hợp lý để có thể chỉ ngay ra lát cắt s-t mà không cần tìm luồng cực đại.

Cách 3: Với một tập $A \subseteq V$ và một đỉnh $v \in V$, ta quan tâm tới giá trị $c(A, \{v\})$, về mặt trực quan, $c(A, \{v\})$ cho biết đỉnh v gắn với A “chặt” tới mức nào.

Lấy u là một đỉnh bất kỳ trong V, đặt $A := \{u\}$. Sau đó cứ chọn đỉnh gắn với A chặt nhất thêm vào A cho tới khi $A = V$ (Dùng kỹ thuật tương tự như thuật toán Prim -PHẦN 4, §9, 9.3). Gọi t là đỉnh được kết nạp cuối cùng trong tiến trình này và s là đỉnh được kết nạp liền trước t.

Định lý 18: Lát cắt $(V \setminus \{t\}, \{t\})$ là lát cắt s-t hẹp nhất.

Chứng minh: Với một lát cắt s-t bất kỳ, ta sẽ chứng minh rằng khả năng thông qua của lát cắt này lớn hơn hay bằng khả năng thông qua của lát cắt $(V \setminus \{t\}, \{t\})$.

Xét một lát cắt s-t bất kỳ ξ . Ta gọi đỉnh v là hoạt tính nếu v và đỉnh được kết nạp liền trước v bị rơi vào hai phần khác nhau của lát cắt ξ . Với mỗi đỉnh v, ta gọi A_v là tập những đỉnh được kết nạp vào A trước đỉnh v. Gọi ξ_v là lát cắt ξ hạn chế trên $A_v \cup \{v\}$ (lát cắt ξ_v dùng đúng cách phân hoạch của lát cắt ξ , nhưng chỉ quan tâm đến tập đỉnh $A_v \cup \{v\}$). Ký hiệu $w(\xi)$ là khả năng thông qua của lát cắt ξ , ký hiệu $w(\xi_v)$ là khả năng thông qua của lát cắt ξ_v .

Bổ đề: Nếu v là đỉnh hoạt tính thì $c(A_v, \{v\}) \leq w(\xi_v)$

Thật vậy, nếu v là đỉnh hoạt tính đầu tiên được kết nạp vào A, khi đó ξ_v sẽ chia tập $A_v \cup \{v\}$ thành hai tập con mà một trong hai tập con đó bằng $\{v\}$ do v là đỉnh hoạt tính đầu tiên. Trong trường hợp này ta có $c(A_v, \{v\}) = w(\xi_v)$. Giả thiết quy nạp rằng bổ đề đúng với u, ta sẽ chứng minh bổ đề cũng đúng với đỉnh hoạt tính v được kết nạp vào A sau u. Thật vậy:

$$\begin{aligned}
 c(A_v, \{v\}) &= c(A_u, \{v\}) + c(A_v \setminus A_u, \{v\}) \quad (\text{Do } A_u \subseteq A_v) \\
 &\leq c(A_u, \{u\}) + c(A_v \setminus A_u, \{v\}) \quad (\text{Do } u \text{ phải gắn với } A_u \text{ chặt hơn hoặc bằng } v) \\
 &\leq w(\xi_u) + c(A_v \setminus A_u, \{v\}) \quad (\text{Giả thiết quy nạp}) \\
 &\leq w(\xi_v) \quad (\text{Do } u \text{ và } v \text{ đều hoạt tính, } A_v \setminus A_u \text{ phải nằm} \\
 &\quad \text{khác phía với } v \text{ trong lát cắt } \xi \Rightarrow \text{Khi tính } w(\xi_v), \text{ tập cạnh có mặt trong phép tính } w(\xi_u) \\
 &\quad \text{cũng có mặt do } A_u \subseteq A_v, \text{ tập cạnh nối từ } \\
 &\quad A_v \setminus A_u \text{ sang } v \text{ cũng có mặt, hai tập cạnh này} \\
 &\quad \text{không giao nhau})
 \end{aligned}$$

Áp dụng bô đè với t là đỉnh hoạt tính, ta có $c(A_t, \{t\}) \leq w(\xi_t) = w(\xi)$. Tức là khả năng thông qua của lát cắt $(V \setminus \{t\}, t)$ nhỏ hơn khả năng thông qua của lát cắt ξ , điều này đúng với mọi ξ , đây là điều phải chứng minh.

Định lý 19: Bài toán tìm lát cắt tổng quát hẹp nhất trên đồ thị có thể giải quyết bằng thuật toán có độ phức tạp $O(n(m+n)\lg n)$, với n và m lần lượt là số cạnh và số đỉnh của đồ thị.


Chứng minh: Với việc áp dụng kỹ thuật gán nhãn tương tự như thuật toán Prim kết hợp với việc sử dụng cấu trúc Heap để chọn ra đỉnh gắn với A chặt nhất, việc xác định s, t và chọn ra lát cắt $s-t$ hẹp nhất có độ phức tạp $O((m+n)\lg n)$. Sau khi ghi nhận lát cắt $s-t$ hẹp nhất này, ta chập s và t lại thành một đỉnh st và lặp lại với đồ thị $G/\{s, t\}$. Tổng cộng có $n - 2$ lần lặp, từ đó suy ra kết quả.

Đây là một bài toán hay, việc chứng minh tính đúng đắn của thuật toán phải dựa trên lý thuyết về luồng cực đại và lát cắt $s-t$ hẹp nhất, nhưng để lập trình giải bài toán lát cắt tổng quát hẹp nhất thì không cần động chạm gì đến các thuật toán tìm luồng cực đại cả.

§11. BÀI TOÁN TÌM BỘ GHÉP CỰC ĐẠI TRÊN ĐỒ THỊ HAI PHÍA

11.1. ĐỒ THỊ HAI PHÍA (BIPARTITE GRAPH)

Các tên gọi đồ thị hai phía một dạng đơn đồ thị vô hướng $G = (V, E)$ mà tập đỉnh của nó có thể chia làm hai tập con X, Y rời nhau sao cho bất kỳ cạnh nào của đồ thị cũng nối một đỉnh của X với một đỉnh thuộc Y . Khi đó người ta còn ký hiệu G là $(X \cup Y, E)$ và gọi một tập (chẳng hạn tập X) là **tập các đỉnh trái** và tập còn lại (chẳng hạn tập Y) là **tập các đỉnh phải** của đồ thị hai phía G . Các đỉnh thuộc X còn gọi là các $X_đỉnh$, các đỉnh thuộc Y gọi là các $Y_đỉnh$.



Hình 84: Đồ thị hai phía

Để kiểm tra một đồ thị liên thông có phải là đồ thị hai phía hay không, ta có thể áp dụng thuật toán sau:

Với một đỉnh v bất kỳ:

```

 $X := \{v\}; Y := \emptyset;$ 
repeat
     $Y := Y \cup K_{\bar{e}}(X);$ 
     $X := X \cup K_{\bar{e}}(Y);$ 
until  $(X \cap Y \neq \emptyset)$  or  $(X \text{ và } Y \text{ là tối đại - không bổ sung được nữa});$ 
if  $X \cap Y \neq \emptyset$  then
    <Không phải đồ thị hai phía>
else
    <Đây là đồ thị hai phía,
    X là tập các đỉnh trái: các đỉnh đến được từ v qua một số chẵn cạnh;
    Y là tập các đỉnh phải: các đỉnh đến được từ v qua một số lẻ cạnh>;

```

Đồ thị hai phía gặp rất nhiều mô hình trong thực tế. Chẳng hạn quan hệ hôn nhân giữa tập những người đàn ông và tập những người đàn bà, việc sinh viên chọn trường, thầy giáo chọn tiết dạy trong thời khoá biểu v.v...

11.2. BÀI TOÁN GHÉP ĐÔI KHÔNG TRỌNG VÀ CÁC KHÁI NIỆM

Cho một đồ thị hai phía $G = (X \cup Y, E)$ ở đây X là tập các đỉnh trái và Y là tập các đỉnh phải của G . $X = \{x[1], x[2], \dots, x[m]\}$, $Y = \{y[1], y[2], \dots, y[n]\}$

Một bộ ghép (matching) của G là một tập hợp các cạnh của G đôi một không có đỉnh chung.

Bài toán ghép đôi (matching problem) là tìm một bộ ghép lớn nhất (nghĩa là có số cạnh lớn nhất) của G

Xét một bộ ghép M của G.

- ❖ Các đỉnh trong M gọi là các đỉnh đã ghép (matched vertices), các đỉnh khác là chưa ghép.
- ❖ Các cạnh trong M gọi là các cạnh đã ghép, các cạnh khác là chưa ghép
- ❖ Nếu định hướng lại các cạnh của đồ thị thành cung, những cạnh chưa ghép được định hướng từ X sang Y, những cạnh đã ghép định hướng từ Y về X. Trên đồ thị định hướng đó: Một đường đi xuất phát từ một X_đỉnh chưa ghép gọi là đường pha, một đường đi từ một X_đỉnh chưa ghép tới một Y_đỉnh chưa ghép gọi là đường mở.

Một cách dễ hiểu, có thể quan niệm như sau:


- ❖ Một đường pha (alternating path) là một đường đi đơn trong G bắt đầu bằng một X_đỉnh chưa ghép, đi theo một cạnh **chưa ghép** sang Y, rồi đến một cạnh **đã ghép** về X, rồi lại đến một cạnh **chưa ghép** sang Y... cứ xen kẽ nhau như vậy.
- ❖ Một đường mở (augmenting path) là một đường pha. Bắt đầu từ một X_đỉnh chưa ghép kết thúc bằng một Y_đỉnh chưa ghép.

Ví dụ: với đồ thị hai phía trong hình Hình 85 và bộ ghép $M = \{(x[1], y[1]), (x[2], y[2])\}$

$x[3]$ và $y[3]$ là những đỉnh chưa ghép, các đỉnh khác là đã ghép

Đường $(x[3], y[2], x[2], y[1])$ là đường pha

Đường $(x[3], y[2], x[2], y[1], x[1], y[3])$ là đường mở.



Hình 85: Đồ thị hai phía và bộ ghép M

11.3. THUẬT TOÁN ĐƯỜNG MỞ

Thuật toán đường mở để tìm một bộ ghép lớn nhất phát biểu như sau:

Bước 1:

Bắt đầu từ một bộ ghép bất kỳ M (thông thường bộ ghép được khởi tạo bằng bộ ghép rỗng hay được tìm bằng các thuật toán tham lam)

Bước 2:

Tìm một đường mở

Bước 3:

- ❖ Nếu bước 2 tìm được đường mở thì mở rộng bộ ghép M: Trên đường mở, loại bỏ những cạnh đã ghép khỏi M và thêm vào M những cạnh chưa ghép. Sau đó lặp lại bước 2.
- ❖ Nếu bước 2 không tìm được đường mở thì thuật toán kết thúc

(Khởi tạo một bộ ghép M);

while (Có đường mở xuất phát từ x tới một đỉnh y chưa ghép $\in Y$) do

 Đọc trên đường mở, xoá bỏ khỏi M các cạnh đã ghép và thêm vào M những cạnh chưa ghép;

 (Sau thao tác này, đỉnh x và y trở thành đã ghép, số cạnh đã ghép tăng lên 1)

Như ví dụ trên, với bộ ghép hai cạnh $M = \{(x[1], y[1]), (x[2], y[2])\}$ và đường mở tìm được gồm các cạnh:

$$(x[3], y[2]) \notin M$$

$$(y[2], x[2]) \in M$$

$$(x[2], y[1]) \notin M$$

$$(y[1], x[1]) \in M$$

$$(x[1], y[3]) \notin M$$

Vậy thì ta sẽ loại đi các cạnh $(y[2], x[2])$ và $(y[1], x[1])$ trong bộ ghép cũ và thêm vào đó các cạnh $(x[3], y[2]), (x[2], y[1]), (x[1], y[3])$ được bộ ghép 3 cạnh.

11.4. CÀI ĐẶT

11.4.1. Biểu diễn đồ thị hai phía

Giả sử đồ thị hai phía $G = (X \cup Y, E)$ có các X _đỉnh ký hiệu là $x[1], x[2], \dots, x[m]$ và các Y _đỉnh ký hiệu là $y[1], y[2], \dots, y[n]$. Ta sẽ biểu diễn đồ thị hai phía này bằng ma trận A cỡ $m \times n$. Trong đó:

$$A[i, j] = \text{TRUE} \Leftrightarrow \text{có cạnh nối đỉnh } x[i] \text{ với đỉnh } y[j].$$

11.4.2. Biểu diễn bộ ghép

Để biểu diễn bộ ghép, ta sử dụng hai mảng: $\text{match}_X[1..m]$ và $\text{match}_Y[1..n]$.

❖ $\text{match}_X[i]$ là chỉ số của đỉnh thuộc tập Y ghép với đỉnh $x[i]$

❖ $\text{match}_Y[j]$ là chỉ số của đỉnh thuộc tập X ghép với đỉnh $y[j]$.

Tức là nếu như cạnh $(x[i], y[j])$ thuộc bộ ghép thì $\text{match}_X[i] = j$ và $\text{match}_Y[j] = i$.

Quy ước rằng:

❖ Nếu như $x[i]$ chưa ghép với đỉnh nào của tập Y thì $\text{match}_X[i] = 0$

❖ Nếu như $y[j]$ chưa ghép với đỉnh nào của tập X thì $\text{match}_Y[j] = 0$.

Suy ra

❖ Thêm một cạnh $(x[i], y[j])$ vào bộ ghép \Leftrightarrow Đặt $\text{match}_X[i] := j$ và $\text{match}_Y[j] := i$;

❖ Loại một cạnh $(x[i], y[j])$ khỏi bộ ghép \Leftrightarrow Đặt $\text{match}_X[i] := 0$ và $\text{match}_Y[j] := 0$;

11.4.3. Tìm đường mở như thế nào.


Vì đường mở bắt đầu từ một $X_{\text{đỉnh}}$ chưa ghép, đi theo một cạnh chưa ghép sang tập Y , rồi theo một đã ghép để về tập X , rồi lại một cạnh chưa ghép sang tập Y ... **cuối cùng là cạnh chưa ghép** tới một $Y_{\text{đỉnh}}$ chưa ghép. Nên có thể thấy ngay rằng độ dài đường mở là lẻ và trên đường mở số cạnh $\in M$ ít hơn số cạnh $\notin M$ là 1 cạnh. Và cũng dễ thấy rằng giải thuật tìm đường mở nên sử dụng thuật toán tìm kiếm theo chiều rộng để đường mở tìm được là đường đi ngắn nhất, giảm bớt công việc cho bước tăng cặp ghép.

Ta khởi tạo một hàng đợi (Queue) ban đầu chứa tất cả các $X_{\text{đỉnh}}$ chưa ghép. Thuật toán tìm kiếm theo chiều rộng làm việc theo nguyên tắc lấy một đỉnh v khỏi Queue và lại đẩy Queue những nối từ v chưa được thăm. Như vậy nếu thăm tới một $Y_{\text{đỉnh}}$ chưa ghép thì tức là ta tìm đường mở kết thúc ở $Y_{\text{đỉnh}}$ chưa ghép đó, quá trình tìm kiếm dừng ngay. Còn nếu ta thăm tới một đỉnh $y[j] \in Y$ đã ghép, dựa vào sự kiện: **từ $y[j]$ chỉ có thể tới được $\text{match}_Y[j]$** theo duy nhất một cạnh đã ghép định hướng ngược từ Y về X , nên ta có thể **đánh dấu thăm $y[j]$, thăm luôn cả $\text{match}_Y[j]$, và đẩy vào Queue phần tử $\text{match}_Y[j] \in X$** (Thăm liền 2 bước).

Input: file văn bản MATCH.INP

- ❖ Dòng 1: chứa hai số m, n ($m, n \leq 1000$) theo thứ tự là số $X_{\text{đỉnh}}$ và số $Y_{\text{đỉnh}}$ cách nhau ít nhất một dấu cách
- ❖ Các dòng tiếp theo, mỗi dòng ghi hai số i, j cách nhau ít nhất một dấu cách thể hiện có cạnh nối hai đỉnh ($x[i], y[j]$).

Output: file văn bản MATCH.OUT, ghi bộ ghép cực đại tìm được



P_4_11_1.PAS * Thuật toán đường mở tìm bộ ghép cực đại

```
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)  
program MatchingProblem;  
const  
  InputFile = 'MATCH.INP';  
  OutputFile = 'MATCH.OUT';  
  max = 1000;  
var  
  m, n: Integer;  
  a: array[1..max, 1..max] of Boolean;  
  matchX, matchY: array[1..max] of Integer;  
  Trace: array[1..max] of Integer;  
  
procedure Enter;
```

```

var
  i, j: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  FillChar(a, SizeOf(a), False);
  ReadLn(f, m, n);
  while not SeekEof(f) do
    begin
      ReadLn(f, i, j);
      a[i, j] := True;
    end;
  Close(f);
end;

procedure Init; {Khởi tạo bộ ghép rỗng}
begin
  FillChar(matchX, SizeOf(matchX), 0);
  FillChar(matchY, SizeOf(matchY), 0);
end;

{Tìm đường mở, nếu thấy trả về một Y_định chưa ghép là đỉnh kết thúc đường mở, nếu không thấy trả về 0}
function FindAugmentingPath: Integer;
var
  Queue: array[1..max] of Integer;
  i, j, Front, Rear: Integer;
begin
  FillChar(Trace, SizeOf(Trace), 0); {Trace[j] = X_định liền trước y[j] trên đường mở}
  Rear := 0; {Khởi tạo hàng đợi rỗng}
  for i := 1 to m do {Đây tất cả những X_định chưa ghép vào hàng đợi}
    if matchX[i] = 0 then
      begin
        Inc(Rear);
        Queue[Rear] := i;
      end;
  {Thuật toán tìm kiếm theo chiều rộng}
  Front := 1;
  while Front <= Rear do
    begin
      i := Queue[Front]; Inc(Front); {Lấy một X_định ra khỏi Queue (x[i])}
      for j := 1 to n do {Xét những Y_định chưa thăm kề với x[i] qua một cạnh chưa ghép}
        if (Trace[j] = 0) and a[i, j] and (matchX[i] <> j) then
          begin {lệnh if trên hơi thừa dk matchX[i] <> j, điều kiện Trace[j] = 0 đã bao hàm luôn điều kiện này rồi}
            Trace[j] := i; {Lưu vết đường đi}
            if matchY[j] = 0 then {Nếu j chưa ghép thì ghi nhận đường mở và thoát ngay}
              begin
                FindAugmentingPath := j;
                Exit;
              end;
            Inc(Rear); {Đây luôn matchY[j] vào hàng đợi}
            Queue[Rear] := matchY[j];
          end;
    end;
  FindAugmentingPath := 0; {Ở trên không Exit được tức là không còn đường mở}
end;

{Nối rộng bộ ghép bằng đường mở kết thúc ở f ∈ Y}
procedure Enlarge(f: Integer);
var
  x, next: Integer;
begin
  repeat
    x := Trace[f];

```

```

next := matchX[x];
matchX[x] := f;
matchY[f] := x;
f := next;
until f = 0;
end;

procedure Solve; {Thuật toán đường mờ}
var
  finish: Integer;
begin
  repeat
    finish := FindAugmentingPath; {Đầu tiên thử tìm một đường mờ}
    if finish <> 0 then Enlarge(finish); {Nếu thấy thì tăng cặp và lặp lại}
  until finish = 0; {Nếu không thấy thì dừng}
end;


procedure PrintResult; {In kết quả}
var
  i, Count: Integer;
  f: Text;
begin
  Assign(f, OutputFile); Rewrite(f);
  WriteLn(f, 'Match: ');
  Count := 0;
  for i := 1 to m do
    if matchX[i] <> 0 then
      begin
        Inc(Count);
        WriteLn(f, Count, ') x[' , i, ', ' - y[' , matchX[i], ', ']);
      end;
  Close(f);
end;

begin
  Enter;
  Init;
  Solve;
  PrintResult;
end.

```

Khảo sát tính đúng đắn của thuật toán cho ta một kết quả khá thú vị:

Nếu ta thêm một đỉnh s và cho thêm m cung từ s tới tất cả những đỉnh của tập X, thêm một đỉnh t và nối thêm n cung từ tất cả các đỉnh của Y tới t. Ta được một mạng với đỉnh phát s và đỉnh thu t.



Hình 86: Mô hình luồng của bài toán tìm bộ ghép cực đại trên đồ thị hai phía

Nếu đặt khả năng thông qua của các cung đều là 1 sau đó tìm luồng cực đại trên mạng thì theo định lý về tính nguyên, luồng dương tìm được trên các cung đều phải là số nguyên (tức là bằng 1 hoặc 0). Khi đó dễ thấy rằng những cung có luồng dương bằng 1 từ tập X tới tập Y sẽ cho ta một bộ ghép lớn nhất. Để chứng minh thuật toán đường mở tìm được bộ ghép lớn nhất sau hữu hạn bước, ta sẽ chứng minh rằng số bộ ghép tìm được bằng thuật toán đường mở sẽ bằng giá trị luồng cực đại nói trên, điều đó cũng rất dễ bởi vì nếu để ý kỹ một chút thì đường mở chẳng qua là đường tăng luồng trên mạng thặng dư mà thôi, ngay cái tên augmenting path đã cho ta biết điều này. Vì vậy thuật toán đường mở ở trường hợp này là một **cách cài đặt hiệu quả trên một dạng đồ thị đặc biệt**, nó làm cho chương trình sáng sủa hơn nhiều so với phương pháp tìm bộ ghép dựa trên bài toán luồng thuần túy.

Người ta đã chứng minh được chi phí thời gian thực hiện giải thuật này trong trường hợp xấu nhất sẽ là $O(n^3)$ đối với đồ thị dày và $O(n(n + m)\log n)$ đối với đồ thị mỏng. Tuy nhiên, cũng giống như thuật toán Ford-Fulkerson, trên thực tế phương pháp này hoạt động rất nhanh.

Bài tập

Bài 1

Có n thợ và m công việc ($n, m \leq 100$). Mỗi thợ cho biết mình có thể làm được những việc nào, hãy phân công các thợ làm các công việc đó sao cho:

- ❖ Mỗi thợ phải làm ít nhất hai việc
- ❖ Một việc chỉ giao cho một thợ thực hiện
- ❖ Số việc thực hiện được là nhiều nhất có thể.

Hướng dẫn: Dựng đồ thị hai phía $G = (X \cup Y, E)$, X tập các thợ và Y là tập các công việc. Sử dụng thuật toán đường mở với cách hiểu đường mở là đường pha xuất phát từ 1 thợ chưa được phân đú hai việc và kết thúc ở một việc chưa được phân công.

Bài 2

Có n thợ và m công việc ($n, m \leq 100$). Mỗi thợ cho biết mình có thể làm được những việc nào, hãy phân công thực hiện các công việc đó sao cho:

- ❖ Mỗi công việc chỉ giao cho một thợ thực hiện
- ❖ Số công việc phân cho người thợ làm nhiều nhất là cực tiêu.

Hướng dẫn: Dựng đồ thị hai phía $G = (X \cup Y, E)$, X là tập các công việc và Y là tập thợ. Với một số nguyên k ($1 \leq k \leq m$), tìm cách phân công thực hiện các công việc sao cho không thợ nào làm quá k việc. Có thể sử dụng thuật toán đường mở với cách hiểu đường mở là đường pha xuất phát từ một việc chưa được phân công và kết thúc ở một thợ chưa làm đú k việc.

Số k sẽ được thử lần lượt từ 1 tới m , xác định số k đầu tiên cho phép tìm ra phép phân công thực hiện hết các việc và in kết quả ra phép phân công tương ứng với số k đó. Có hai điều quan trọng phải lưu ý: Thứ nhất là phép thử với $k = k_0$ có thể tận dụng kết quả của phép phân công với $k = k_0 - 1$ chứ không cần thực hiện lại từ đầu. Thứ hai là việc “cải tiến” bằng áp dụng thuật toán tìm kiếm nhị phân để chỉ ra số k nhỏ nhất thỏa mãn yêu cầu phân công toàn

bộ các công việc sẽ không giúp ích gì mà sẽ chỉ làm chương trình chậm đi, lý do là bởi mỗi lần tăng cặp dựa trên đường mở thì số việc được phân công cũng chỉ tăng lên 1 mà thôi.

Bài 3

Xem lại bài toán Minimum Path Cover ở §10. Cài đặt lại theo cách bớt còng kẽm hơn bằng cách sử dụng những kỹ thuật đã học được trong bài toán tìm bộ ghép cực đại.

Bài 4

Cho đồ thị hai phía $G = (X \cup Y, E)$. Hãy chỉ ra một tập S gồm ít nhất các đỉnh sao cho mỗi cạnh $\in E$ đều liên thuộc với ít nhất một đỉnh thuộc S .

Hướng dẫn: Có thể đưa về mô hình bài toán luồng với khả năng thông qua của cả các cung và các đỉnh: Đưa một đỉnh phát giả s nối tới mọi đỉnh $\in X$, một đỉnh thu giả t nối từ mọi đỉnh $\in Y$. Các cạnh trong E được định chiều từ X sang Y , khả năng thông qua của các cung được đặt bằng $+\infty$. Khả năng thông qua của các đỉnh $\in X \cup Y$ đặt bằng 1. Sau đó tìm luồng cực đại từ s tới t và lát cắt hẹp nhất, lát cắt này chắc chắn cắt tại các đỉnh, những đỉnh bị cắt sẽ được chọn vào tập S . Tuy nhiên, bằng một số suy luận, ta có thể đưa về mô hình bài toán bộ ghép cực đại để dễ dàng hơn trong việc cài đặt bằng phương pháp sau:

- ❖ Tìm bộ ghép cực đại trên G
- ❖ Khi tìm xong bộ ghép, tức là thủ tục tìm đường mở không tìm ra đường mở, ta xác định được:
 - Tập $Y^* = \{\text{Tập các } Y\text{-đỉnh đi đến được từ một } X\text{-đỉnh chưa ghép qua một đường pha}\}$
 - Tập $X^* = \{x \in X | x \text{ đã ghép và đỉnh ghép với } x \notin Y^*\}$
- ❖ Đặt $S = X^* \cup Y^*$

§12. BÀI TOÁN TÌM BỘ GHÉP CỰC ĐẠI VỚI TRỌNG SỐ CỰC TIỀU TRÊN ĐỒ THỊ HAI PHÍA - THUẬT TOÁN HUNGARI

12.1. BÀI TOÁN PHÂN CÔNG

Đây là một dạng bài toán phát biểu như sau: Có m người (đánh số 1, 2, ..., m) và n công việc (đánh số 1, 2, ..., n), mỗi người có khả năng thực hiện một số công việc nào đó. Để giao cho người i thực hiện công việc j cần một chi phí là $c[i, j] \geq 0$. Cần phân cho mỗi thợ một việc và mỗi việc chỉ do một thợ thực hiện sao cho số công việc có thể thực hiện được là nhiều nhất và nếu có ≥ 2 phương án đều thực hiện được nhiều công việc nhất thì chỉ ra phương án chi phí ít nhất.

Dựng đồ thị hai phía $G = (X \cup Y, E)$ với X là tập m người, Y là tập n việc và $(u, v) \in E$ với trọng số $c[u, v]$ nếu như người u làm được công việc v . Bài toán đưa về tìm bộ ghép nhiều cạnh nhất của G có trọng số nhỏ nhất.

Gọi $k = \max(m, n)$. Bổ sung vào tập X và Y một số đỉnh giả đê $|X| = |Y| = k$.

Gọi M là một số dương đủ lớn hơn chi phí của mọi phép phân công có thể. Với mỗi cặp đỉnh (u, v) : $u \in X$ và $v \in Y$. Nếu $(u, v) \notin E$ thì ta bổ sung cạnh (u, v) vào E với trọng số là M .

Khi đó ta được G là một đồ thị hai phía đầy đủ (Đồ thị hai phía mà giữa một đỉnh bất kỳ của X và một đỉnh bất kỳ của Y đều có cạnh nối). Và nếu như ta tìm được bộ ghép đầy đủ k cạnh mang trọng số nhỏ nhất thì ta chỉ cần loại bỏ khỏi bộ ghép đó những cạnh mang trọng số M vừa thêm vào thì sẽ được kế hoạch phân công 1 người \leftrightarrow 1 việc cần tìm. Điều này dễ hiểu bởi bộ ghép đầy đủ mang trọng số nhỏ nhất tức là phải ít cạnh trọng số M nhất, tức là số phép phân công là nhiều nhất, và tất nhiên trong số các phương án ghép ít cạnh trọng số M nhất thì đây là phương án trọng số nhỏ nhất, tức là tổng chi phí trên các phép phân công là ít nhất.

12.2. PHÂN TÍCH

Input: Đồ thị hai phía đầy đủ $G = (X \cup Y, E)$; $X = \{x[1], \dots, x[k]\}$, $Y = \{y[1], \dots, y[k]\}$. Được cho bởi ma trận vuông C cỡ $k \times k$, $c[i, j] =$ trọng số cạnh nối đỉnh $x[i]$ với $y[j]$. Giả thiết $c[i, j] \geq 0 (\forall i, j)$

Output Bộ ghép đầy đủ trọng số nhỏ nhất.

Hai định lý sau đây tuy rất đơn giản nhưng là những định lý quan trọng tạo cơ sở cho thuật toán sẽ trình bày:

Định lý 1: Loại bỏ khỏi G những cạnh trọng số > 0 . Nếu những cạnh trọng số 0 còn lại tạo ra bộ ghép k cạnh trong G thì đây là bộ ghép cần tìm.

Chứng minh: Theo giả thiết, các cạnh của G mang trọng số không âm nên bất kỳ bộ ghép nào trong G cũng có trọng số không âm, mà bộ ghép ở trên mang trọng số 0, nên tất nhiên đó là bộ ghép đầy đủ trọng số nhỏ nhất.

Định lý 2: Với đỉnh $x[i]$, nếu ta cộng thêm một số Δ (đương hay âm) vào tất cả những cạnh liên thuộc với $x[i]$ (tương đương với việc cộng thêm Δ vào tất cả các phần tử thuộc hàng i của ma trận C) thì không ảnh hưởng tới bộ ghép đầy đủ trọng số nhỏ nhất.

Chứng minh: Với một bộ ghép đầy đủ bất kỳ thì có một và chỉ một cạnh ghép với $x[i]$. Nên việc cộng thêm Δ vào tất cả các cạnh liên thuộc với $x[i]$ sẽ làm tăng trọng số bộ ghép đó lên Δ . Vì vậy nếu như ban đầu, M là bộ ghép đầy đủ trọng số nhỏ nhất thì sau thao tác trên, M vẫn là bộ ghép đầy đủ trọng số nhỏ nhất.

Hệ quả: Với đỉnh $y[j]$, nếu ta cộng thêm một số Δ (dương hay âm) vào tất cả những cạnh liên thuộc với $y[j]$ (tương đương với việc cộng thêm Δ vào tất cả các phần tử thuộc cột j của ma trận C) thì không ảnh hưởng tới bộ ghép đầy đủ trọng số nhỏ nhất.

Tùy đây có thể nhận ra tư tưởng của thuật toán: Từ đồ thị G , ta tìm chiến lược cộng / trừ một cách hợp lý trọng số của các cạnh liên thuộc với từng đỉnh để được một đồ thị mới vẫn có các cạnh trọng số không âm, mà các cạnh trọng số 0 của đồ thị mới đó chứa một bộ ghép đầy đủ k cạnh.

Ví dụ: Biến đổi ma trận trọng số của đồ thị hai phía 3 đỉnh trái, 3 đỉnh phải.

$$\begin{array}{ccc} & \xrightarrow{\quad\quad} & \\ \left(\begin{array}{ccc} 0 & 0 & 0 \\ 0 & 1 & 7 \\ 0 & 8 & 9 \end{array} \right) & \xrightarrow{\quad\quad} & \left(\begin{array}{ccc} 1 & 0 & 0 \\ 0 & 0 & 6 \\ 0 & 7 & 8 \end{array} \right) \xrightarrow{\quad\quad} & x[1]-y[3] \\ -1 \longleftarrow & & & x[2]-y[2] \\ -1 \longleftarrow & & & x[3]-y[1] \\ +1 & & & \end{array}$$

Hình 87: Phép xoay trọng số cạnh

12.3. THUẬT TOÁN

12.3.1. Các khái niệm:

Để cho gọn, ta gọi những cạnh trong số 0 của G là những 0 cạnh.

Xét một bộ ghép M chỉ gồm những 0 cạnh.

- ❖ Những đỉnh $\in M$ gọi là những đỉnh đã ghép, những đỉnh còn lại gọi là những đỉnh chưa ghép.
 - ❖ Những 0_cạnh $\in M$ gọi là những 0_cạnh đã ghép, những 0_cạnh còn lại là những 0_cạnh chưa ghép.
 - ❖ Nếu ta định hướng lại các 0_cạnh theo cách: Những 0_cạnh chưa ghép cho hướng từ tập X sang tập Y, những 0_cạnh đã ghép cho hướng từ tập Y về tập X. Khi đó:
 - ❖ Đường pha (Alternating Path) là một đường đi cơ bản xuất phát từ một X_đỉnh chưa ghép đi theo các 0_cạnh đã định hướng ở trên. Như vậy dọc trên đường pha, các 0_cạnh chưa ghép và những 0_cạnh đã ghép xen kẽ nhau. Vì đường pha chỉ là đường đi cơ bản trên đồ

thì định hướng nên việc xác định những đỉnh nào có thể đến được từ $x \in X$ bằng một đường pha có thể sử dụng các thuật toán tìm kiếm trên đồ thị (BFS hoặc DFS). Những đỉnh và những cạnh được duyệt qua tạo thành một cây pha gốc x

- ❖ Một đường mở (Augmenting Path) là một đường pha đi từ một $X_{\text{đỉnh chưa ghép}}$ tới một $Y_{\text{đỉnh chưa ghép}}$.

Như vậy:

- ❖ Đường đi trực tiếp từ một $X_{\text{đỉnh chưa ghép}}$ tới một $Y_{\text{đỉnh chưa ghép}}$ qua một $0_{\text{cạnh chưa ghép}}$ cũng là một đường mở.
- ❖ Dọc trên đường mở, số $0_{\text{cạnh chưa ghép}}$ nhiều hơn số $0_{\text{cạnh đã ghép}}$ đúng 1 cạnh.

12.3.2. Thuật toán Hungari

Bước 1: Khởi tạo:

Một bộ ghép $M := \emptyset$

Bước 2: Với mọi đỉnh $x^* \in X$, ta tìm cách ghép x^* :

Bắt đầu từ đỉnh x^* , thử tìm đường mở bắt đầu ở x^* bằng thuật toán tìm kiếm trên đồ thị. Có hai khả năng có thể xảy ra:

- ❖ Hoặc tìm được đường mở thì dọc theo đường mở, ta loại bỏ những cạnh đã ghép khỏi M và thêm vào M những cạnh chưa ghép, ta được một **bộ ghép mới nhiều hơn bộ ghép cũ 1 cạnh và đỉnh x^* trở thành đã ghép**.
- ❖ Hoặc không tìm được đường mở thì có thể xác định được:

$\text{VisitedX} = \{\text{Tập những } X_{\text{đỉnh}} \text{ có thể đến được từ } x^* \text{ bằng một đường pha}\}$

$\text{VisitedY} = \{\text{Tập những } Y_{\text{đỉnh}} \text{ có thể đến được từ } x^* \text{ bằng một đường pha}\}$

Gọi Δ là trọng số nhỏ nhất của các cạnh nối giữa một đỉnh thuộc VisitedX với một đỉnh không thuộc VisitedY . Để thấy $\Delta > 0$ bởi nếu $\Delta = 0$ thì tồn tại một $0_{\text{cạnh}} (x, y)$ với $x \in \text{VisitedX}$ và $y \notin \text{VisitedY}$. Vì x^* đến được x bằng một đường pha và (x, y) là một $0_{\text{cạnh}}$ nên x^* cũng đến được y bằng một đường pha, dẫn tới $y \in \text{VisitedY}$, điều này vô lý.

Biến đổi đồ thị G : Với $\forall x \in \text{VisitedX}$, trừ Δ vào trọng số những cạnh liên thuộc với x ,
Với $\forall y \in \text{VisitedY}$, cộng Δ vào trọng số những cạnh liên thuộc với y .

Lặp lại thủ tục tìm kiếm trên đồ thị thử tìm đường mở xuất phát ở x^* cho tới khi tìm ra đường mở.

Bước 3: Sau bước 2 thì mọi $X_{\text{đỉnh}}$ đều được ghép, in kết quả về bộ ghép tìm được.

Mô hình cài đặt của thuật toán có thể viết như sau:

```

M := ∅;
for (x* ∈ X) do
begin
repeat
    (Tìm đường mở xuất phát ở x*);
    if (Không tìm thấy đường mở) then (Biến đổi đồ thị G: Chọn Δ := ...);

```


```


until {Tìm thấy đường mở};
  (Đọc theo đường mở, loại bỏ những cạnh đã ghép khỏi M và thêm vào M những cạnh chưa ghép);
end;
{Output: M là bộ ghép cần tìm};

```

Ví dụ minh họa:

Để không bị rối hình, ta hiểu những cạnh không ghi trọng số là những 0_cạnh, những cạnh không vẽ mang trọng số rất lớn trong trường hợp này không cần thiết phải tính đến. Những cạnh nét đậm là những cạnh đã ghép, những cạnh nét thanh là những cạnh chưa ghép.





Hình 88: Thuật toán Hungari

Để ý rằng nếu như không tìm thấy đường mở xuất phát ở x^* thì quá trình tìm kiếm trên đồ thị sẽ cho ta một cây pha gốc x^* . Giá trị xoay Δ thực chất là trọng số nhỏ nhất của cạnh nối một $X_{\text{đỉnh}}$ trong cây pha với một $Y_{\text{đỉnh}}$ ngoài cây pha (cạnh ngoài). Việc trừ Δ vào những cạnh liên thuộc với $X_{\text{đỉnh}}$ trong cây pha và cộng Δ vào những cạnh liên thuộc với $Y_{\text{đỉnh}}$ trong cây pha sẽ làm cho cạnh ngoài nối trên trở thành 0_cạnh, các cạnh khác vẫn có trọng số ≥ 0 . Nhưng quan trọng hơn là **tất cả những cạnh trong cây pha vẫn cứ là 0_cạnh**. Điều đó đảm bảo cho quá trình tìm kiếm trên đồ thị lần sau sẽ xây dựng được cây pha mới lớn hơn cây pha cũ. Vì tập các $Y_{\text{đỉnh}}$ đã ghép là hữu hạn nên sau không quá k bước, cây pha sẽ quét tới một $Y_{\text{đỉnh}}$ chưa ghép, tức là tìm ra đường mở

12.3.3. Phương pháp đối ngẫu Kuhn-Munkres

Phương pháp Kuhn-Munkres đi tìm hai dãy số $Fx[1..k]$ và $Fy[1..k]$ thoả mãn:

- ❖ $c[i, j] - Fx[i] - Fy[j] \geq 0$
- ❖ Tập các cạnh $(x[i], y[j])$ thoả mãn $c[i, j] - Fx[i] - Fy[j] = 0$ chứa trọn một bộ ghép đầy đủ k cạnh, đây chính là bộ ghép cần tìm.

Rõ ràng nếu tìm được hai dãy số thoả mãn trên thì ta chỉ việc thực hiện hai thao tác:

- ❖ Với mỗi đỉnh $x[i]$, trừ tất cả trọng số của những cạnh liên thuộc với $x[i]$ đi $Fx[i]$
- ❖ Với mỗi đỉnh $y[j]$, trừ tất cả trọng số của những cạnh liên thuộc với $y[j]$ đi $Fy[j]$

(Hai thao tác này tương đương với việc trừ tất cả trọng số của các cạnh $(x[i], y[j])$ đi một lượng $Fx[i] + Fy[j]$ tức là $c[i, j] := c[i, j] - Fx[i] - Fy[j]$)

Thì dễ thấy đồ thị mới tạo thành sẽ gồm có các cạnh trọng số không âm và những 0_cạnh của đồ thị chứa trọn một bộ ghép đầy đủ.

	1	2	3	4	
1	0	0	M	M	$Fx[1] = 2$
2	0	M	M	2	$Fx[2] = 2$
3	M	1	0	M	$Fx[3] = 3$
4	M	M	0	9	$Fx[4] = 3$
	$Fy[1] = -2$	$Fy[2] = -2$	$Fy[3] = -3$	$Fy[4] = 0$	

(Có nhiều phương án khác: $Fx = (0, 0, 1, 1)$; $Fy = (0, 0, -1, 2)$ cũng đúng)

Vậy phương pháp Kuhn-Munkres đưa việc biến đổi đồ thị G (biến đổi ma trận C) về việc biến đổi hay dãy số Fx và Fy . Việc trừ Δ vào trọng số tất cả những cạnh liên thuộc với $x[i]$ tương đương với việc tăng $Fx[i]$ lên Δ . Việc cộng Δ vào trọng số tất cả những cạnh liên thuộc với $y[j]$ tương đương với giảm $Fy[j]$ đi Δ . Khi cần biết trọng số cạnh $(x[i], y[j])$ là bao nhiêu sau các bước biến đổi, thay vì viết $c[i, j]$, ta viết $c[i, j] - Fx[i] - Fy[j]$.

Sơ đồ cài đặt phương pháp Kuhn-Munkres có thể viết như sau:

Bước 1: Khởi tạo:

- ❖ $M := \emptyset$;
- ❖ Việc khởi tạo các Fx , Fy có thể có nhiều cách miễn sao $c[i, j] - Fx[i] - Fy[j] \geq 0$, đơn giản nhất có thể đặt tất cả các $Fx[.]$ và $Fy[.]$ bằng 0

Bước 2: Với mọi đỉnh $x^* \in X$, ta tìm cách ghép x^* như sau:

Bắt đầu từ đỉnh x^* , thử tìm đường mở bắt đầu ở x^* bằng thuật toán tìm kiếm trên đồ thị (BFS hoặc DFS). Có hai khả năng xảy ra:

- ❖ Hoặc tìm được đường mở thì đọc theo đường mở, ta loại bỏ những cạnh đã ghép khỏi M và thêm vào M những cạnh chưa ghép.
- ❖ Hoặc không tìm được đường mở thì xác định được:

$$\text{VisitedX} = \{\text{Tập những } X\text{-đỉnh có thể đến được từ } x^* \text{ bằng một đường pha}\}$$

$$\text{VisitedY} = \{\text{Tập những } Y\text{-đỉnh có thể đến được từ } x^* \text{ bằng một đường pha}\}$$

$$\text{Đặt } \Delta := \min \{c[i, j] - Fx[i] - Fy[j] \mid \forall x[i] \in \text{VisitedX}; \forall y[j] \notin \text{VisitedY}\}$$

Với $\forall x[i] \in \text{VisitedX}$: đặt $Fx[i] := Fx[i] + \Delta$;

Với $\forall y[j] \in \text{VisitedY}$: đặt $Fy[j] := Fy[j] - \Delta$;

Lặp lại thủ tục tìm đường mở xuất phát tại x^* cho tới khi tìm ra đường mở.

Bước 3:

Lưu ý rằng bước 2 luôn tìm ra đường mở vì đồ thị đã được làm cho trở nên cân bằng ($|X|=|Y|$) và đầy đủ, ta chỉ việc trả về đường mở tìm được.

Đáng lưu ý ở phương pháp Kuhn-Munkres là phương pháp này không làm thay đổi ma trận C ban đầu. Điều đó thực sự hữu ích trong trường hợp trọng số của cạnh $(x[i], y[j])$ không được

cho một cách tường minh bằng giá trị $c[i, j]$ mà lại cho bằng hàm $c(i, j)$: trong trường hợp này, việc trừ hàng/cộng cột trực tiếp trên ma trận chi phí C là không thể thực hiện được.

12.3.4. Cài đặt

a) Biểu diễn bộ ghép

Để biểu diễn bộ ghép, ta sử dụng hai mảng: $\text{matchX}[1..k]$ và $\text{matchY}[1..k]$.

- ❖ $\text{matchX}[i]$ là chỉ số của đỉnh thuộc tập Y ghép với đỉnh $x[i]$
- ❖ $\text{matchY}[j]$ là chỉ số của đỉnh thuộc tập X ghép với đỉnh $y[j]$.

Tức là nếu như cạnh $(x[i], y[j])$ thuộc bộ ghép thì $\text{matchX}[i] = j$ và $\text{matchY}[j] = i$.

Quy ước:

- ❖ Nếu như $x[i]$ chưa ghép với đỉnh nào của tập Y thì $\text{matchX}[i] = 0$
- ❖ Nếu như $y[j]$ chưa ghép với đỉnh nào của tập X thì $\text{matchY}[j] = 0$

Suy ra:

- ❖ Thêm một cạnh $(x[i], y[j])$ vào bộ ghép \Leftrightarrow Đặt $\text{matchX}[i] := j$ và $\text{matchY}[j] := i$;
- ❖ Loại một cạnh $(x[i], y[j])$ khỏi bộ ghép \Leftrightarrow Đặt $\text{matchX}[i] := 0$ và $\text{matchY}[j] := 0$;

b) Tìm đường mở như thế nào


Ta sẽ tìm đường mở và xây dựng hai tập VisitedX và VisitedY bằng thuật toán tìm kiếm theo chiều rộng, chỉ xét những 0_cạnh định hướng như đã nói trong phần đầu:

Khởi tạo một hàng đợi (Queue) ban đầu chỉ có một đỉnh x^* . Thuật toán tìm kiếm theo chiều rộng làm việc theo nguyên tắc lấy một đỉnh v khỏi Queue và lại đẩy Queue những nối từ v chưa được thăm. Như vậy nếu thăm tới một $Y_{\text{đỉnh}}$ chưa ghép thì tức là ta tìm đường mở kết thúc ở $Y_{\text{đỉnh}}$ chưa ghép đó, quá trình tìm kiếm dừng ngay. Còn nếu ta thăm tới một đỉnh $y[j] \in Y$ đã ghép, dựa vào sự kiện: từ $y[j]$ chỉ có thể tới được $\text{matchY}[j]$ theo duy nhất một 0_cạnh định hướng, nên ta có thể đánh dấu thăm $y[j]$, thăm luôn cả $\text{matchY}[j]$, và đẩy vào Queue phần tử $\text{matchY}[j] \in X$.

Input: file văn bản ASSIGN.INP

- ❖ Dòng 1: Ghi hai số m, n theo thứ tự là số thợ và số việc cách nhau 1 dấu cách ($m, n \leq 1000$)
- ❖ Các dòng tiếp theo, mỗi dòng ghi ba số i, j, $c[i, j]$ cách nhau 1 dấu cách thể hiện thợ i làm được việc j và chi phí để làm là $c[i, j]$ ($1 \leq i \leq m; 1 \leq j \leq n; 0 \leq c[i, j] \leq 1000$).

Output: file văn bản ASSIGN.OUT, mô tả phép phân công tối ưu tìm được.



ASSIGN.INP	ASSIGN.OUT
5 6	Optimal assignment:
1 1 0	1) x[1] - y[1] 0
1 2 0	2) x[2] - y[4] 2
2 1 0	3) x[3] - y[2] 1
2 4 2	4) x[4] - y[3] 0
3 2 1	
3 3 0	
4 3 0	
4 4 9	
5 4 19	

Cost: 3

P_4_12_1.PAS * Thuật toán Hungari

```

{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program AssignmentProblemSolve;
const
  InputFile = 'ASSIGN.INP';
  OutputFile = 'ASSIGN.OUT';
  max = 1000;
  maxEC = 1000
  maxC = max * maxEC + 1;
var
  c: array[1..max, 1..max] of Integer;
  Fx, Fy, matchX, matchY, Trace: array[1..max] of Integer;
  m, n, k, start, finish: Integer;

procedure Enter; {Nhập dữ liệu}
var
  i, j: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, m, n);
  if m > n then k := m else k := n;
  for i := 1 to k do
    for j := 1 to k do c[i, j] := maxC;
  while not SeekEof(f) do ReadLn(f, i, j, c[i, j]);
  Close(f);
end;

procedure Init; {Khởi tạo bộ ghép rỗng và các giá trị Fx[.], Fy[.]}
begin
  FillChar(matchX, SizeOf(matchX), 0);
  FillChar(matchY, SizeOf(matchY), 0);
  FillChar(Fx, SizeOf(Fx), 0);
  FillChar(Fy, SizeOf(Fy), 0);
end;

function GetC(i, j: Integer): Integer; {Hàm trả về trọng số cạnh (x[i], y[j])}
begin
  GetC := c[i, j] - Fx[i] - Fy[j];
end;

procedure FindAugmentingPath; {Thủ tục tìm đường mở xuất phát ở x[start]}
var
  Queue: array[1..max] of Integer; {Hàng đợi dùng cho BFS, chỉ chứa chỉ số các đỉnh ∈ X}
  i, j, Front, Rear: Integer;

```

```

begin
  FillChar(Trace, SizeOf(Trace), 0);
  Queue[1] := start;
  Front := 1; Rear := 1;
  repeat
    i := Queue[Front]; Inc(Front); {Lấy i ra khỏi Queue, xét x[i]}
    for j := 1 to k do
      if (Trace[j] = 0) and (GetC(i, j) = 0) then {Nếu y[j] chưa thăm và kè với x[i] qua 0_cạnh}
        begin
          Trace[j] := i; {Lưu vết đường đi}
          if matchY[j] = 0 then {Nếu y[j] đã ghép thì ghi nhận và thoát ngay}
            begin
              finish := j;
              Exit;
            end;
          Inc(Rear); Queue[Rear] := matchY[j]; {Không thì đẩy matchY[j] vào Queue, chờ duyệt tiếp}
        end;
    until Front > Rear;
end;

procedure SubX_AddY; {Phép xoay trọng số cạnh}
var
  i, j, t, Delta: Integer;
  VisitedX, VisitedY: set of Byte;
begin
  {Trước hết tìm hai tập VisitedX và VisitedY chứa chỉ số các đỉnh đến được từ x[start] qua một đường pha}
  VisitedX := [start];
  VisitedY := [];
  for j := 1 to k do
    if Trace[j] <> 0 then
      begin
        Include(VisitedX, matchY[j]);
        Include(VisitedY, j);
      end;
  {Tính Delta := min(GetC(i, j)|i ∈ VisitedX và j ∉ VisitedY)}
  Delta := maxC;
  for i := 1 to k do
    if i in VisitedX then
      for j := 1 to k do
        if not (j in VisitedY) and (GetC(i, j) < Delta) then
          Delta := GetC(i, j);
  {Xoay}
  for t := 1 to k do
    begin
      if t in VisitedX then Fx[t] := Fx[t] + Delta;
      if t in VisitedY then Fy[t] := Fy[t] - Delta;
    end;
end;

procedure Enlarge; {Nối rộng bộ ghép bằng đường mờ kết thúc tại y[finish]}
var
  i, next: Integer;
begin
  repeat
    i := Trace[finish];
    next := matchX[i];
    matchX[i] := finish;
    matchY[finish] := i;
    finish := Next;
  until finish = 0; {finish = 0 ⇔ i = start}
end;

procedure Solve; {Thuật toán Hungari}

```

```

var
  i: Integer;
begin
  for i := 1 to k do
    begin
      start := i; finish := 0;
      repeat {Tìm cách ghép x[start]}
        FindAugmentingPath;
        if finish = 0 then SubX_AddY; {Nếu không tìm ra đường mờ xuất phát từ x[start] thì xoay các trọng số cạnh}
        until finish <> 0;
      Enlarge; {Khi đã tìm ra đường mờ thì chỉ cần tăng cắp theo đường mờ}
    end;
end;

procedure Result; {In kết quả}
var
  i, j, Count, W: Integer;
  f: Text;
begin
  Assign(f, OutputFile); Rewrite(f);
  WriteLn(f, 'Optimal assignment:');
  W := 0; Count := 0;
  for i := 1 to m do
    begin
      j := matchX[i];
      if c[i, j] < maxC then {Chi in quan tâm tới những cạnh trọng số < maxC}
        begin
          Inc(Count);
          WriteLn(f, Count:3, ' ', i, ' - ', j, ' ', c[i, j]);
          W := W + c[i, j];
        end;
    end;
  WriteLn(f, 'Cost: ', W);
  Close(f);
end;

begin
  Enter;
  Init;
  Solve;
  Result;
end.

```

Nhận xét:

- ❖ Có thể giải quyết bài toán phân công nếu như ma trận chi phí C có phần tử âm bằng cách sửa lại một chút trong thủ tục khởi tạo (Init): Với $\forall i$, thay vì gán $Fx[i] := 0$, ta gán $Fx[i] :=$ giá trị nhỏ nhất trên hàng i của ma trận C, khi đó sẽ đảm bảo $c[i, j] - Fx[i] - Fy[j] \geq 0$ ($\forall i, j$).
- ❖ Sau khi kết thúc thuật toán, tổng tất cả các phần tử ở hai dãy Fx, Fy bằng trọng số cực tiểu của bộ ghép đầy đủ tìm được trên đồ thị ban đầu.
- ❖ Một vấn đề nữa phải hết sức cẩn thận trong việc ước lượng độ lớn của các phần tử $Fx[.]$ và $Fy[.]$ khi khai báo mảng, các giá trị này có thể lớn hơn rất nhiều lần so với giá trị lớn nhất của các $c[i, j]$. Hãy tự tìm ví dụ để giải thích tại sao.

12.4. BÀI TOÁN TÌM BỘ GHÉP CỰC ĐẠI VỚI TRỌNG SỐ CỰC ĐẠI TRÊN ĐỒ THỊ HAI PHÍA

Bài toán tìm bộ ghép cực đại với trọng số cực đại cũng có thể giải nhờ phương pháp Hungari bằng cách đổi dấu tất cả các phần tử ma trận chi phí.

Khi cài đặt, ta có thể sửa lại đôi chút trong chương trình trên để giải bài toán tìm bộ ghép cực đại với trọng số cực đại mà không cần đổi dấu trọng số. Cụ thể như sau:

Bước 1: Khởi tạo:

- ❖ $M := \emptyset$;
- ❖ Khởi tạo hai dãy Fx và Fy thỏa mãn: $\forall i, j: Fx[i] + Fy[j] \geq c[i, j]$; chặng hạn ta có thể đặt $Fx[i] :=$ Phần tử lớn nhất trên dòng i của ma trận C và đặt các $Fy[j] := 0$.

Bước 2: Với mọi đỉnh $x^* \in X$, ta tìm cách ghép x^* :

Bắt đầu từ đỉnh x^* , thử tìm đường mở bắt đầu ở x^* . Có hai khả năng xảy ra:

- ❖ Hoặc tìm được đường mở thì đọc theo đường mở, ta loại bỏ những cạnh đã ghép khỏi M và thêm vào M những cạnh chưa ghép.
- ❖ Hoặc không tìm được đường mở thì xác định được:

$$\text{VisitedX} = \{\text{Tập những } X\text{-đỉnh có thể đến được từ } x^* \text{ bằng một đường pha}\}$$

$$\text{VisitedY} = \{\text{Tập những } Y\text{-đỉnh có thể đến được từ } x^* \text{ bằng một đường pha}\}$$

$$\text{Đặt } \Delta := \min \{Fx[i] + Fy[j] - c[i, j] \mid \forall x[i] \in \text{VisitedX}, \forall y[j] \notin \text{VisitedY}\}$$

Xoay trọng số cạnh:

$$\text{Với } \forall x[i] \in \text{VisitedX}: Fx[i] := Fx[i] - \Delta;$$

$$\text{Với } \forall y[j] \in \text{VisitedY}: Fy[j] := Fy[j] + \Delta;$$

Lặp lại thủ tục tìm đường mở xuất phát tại x^* cho tới khi tìm ra đường mở.

Bước 3: Sau bước 2 thì mọi X -đỉnh đều đã ghép, ta được một bộ ghép đầy đủ k cạnh với trọng số lớn nhất.

Dễ dàng chứng minh được tính đúng đắn của phương pháp, bởi nếu ta đặt:

$$c'[i, j] = -c[i, j]; F'x[i] := -Fx[i]; F'y[j] = -Fy[j].$$

Thì bài toán trở thành tìm cặp ghép đầy đủ trọng số cực tiểu trên đồ thị hai phía với ma trận trọng số $c'[1..k, 1..k]$. Bài toán này được giải quyết bằng cách tính hai dãy đối ngẫu $F'x$ và $F'y$. Từ đó bằng những biến đổi đại số cơ bản, ta có thể kiểm chứng được tính tương đương giữa các bước của phương pháp nêu trên với các bước của phương pháp Kuhn-Munkres ở mục trước.

12.5. NÂNG CẤP


Dựa vào mô hình cài đặt thuật toán Kuhn-Munkres ở trên, ta có thể đánh giá về độ phức tạp tính toán lý thuyết của cách cài đặt này:

Thuật toán tìm kiếm theo chiều rộng được sử dụng để tìm đường mở có độ phức tạp $O(k^2)$, mỗi lần xoay trọng số cạnh mất một chi phí thời gian cỡ $O(k^2)$. Vậy mỗi lần tăng gấp, cần tối đa k lần dò đường và k lần xoay trọng số cạnh, mất một chi phí thời gian cỡ $O(k^3)$. Thuật toán cần k lần tăng gấp nên độ phức tạp tính toán trên lý thuyết của phương pháp này cỡ $O(k^4)$.

Có thể cải tiến mô hình cài đặt để được một thuật toán với độ phức tạp $O(k^3)$ dựa trên những nhận xét sau:

12.5.1. Nhận xét 1

Quá trình tìm kiếm theo chiều rộng bắt đầu từ một đỉnh x^* chưa ghép cho ta một cây pha gốc x^* . Nếu tìm được đường mở thì dừng lại và tăng cấp ngay, nếu không thì xoay trọng số cạnh và bắt đầu tìm kiếm lại để được một cây pha mới lớn hơn cây pha cũ (Hình 89):



Hình 89: Cây pha “mọc” lớn hơn sau mỗi lần xoay trong số cạnh và tìm đường

12.5.2. Nhận xét 2

Việc xác định trọng số nhỏ nhất của cạnh nối một X_đỉnh trong cây pha với một Y_đỉnh ngoài cây pha có thể kết hợp ngay trong bước dựng cây pha mà không làm tăng cấp phức tạp tính toán. Để thực hiện điều này, ta sử dụng kỹ thuật như trong thuật toán Prim:

Với mọi $y[j] \in Y$, gọi $d[j] :=$ khoảng cách từ $y[j]$ đến cây pha gốc x^* . Ban đầu $d[j]$ được khởi tạo bằng trọng số cạnh $(x^*, y[j])$ (cây pha ban đầu chỉ có đúng một đỉnh x^*).

Trong bước tìm đường bằng BFS, mỗi lần rút một đỉnh $x[i]$ ra khỏi Queue, ta xét những đỉnh $y[j] \in Y$ chưa thăm và đặt lại $d[j]_{\text{mới}} := \min(d[j]_{\text{cũ}}, \text{trọng số cạnh } (x[i], y[j]))$ sau đó mới kiểm tra xem $(x[i], y[j])$ có phải là 0_cạnh hay không để tiếp tục các thao tác như trước. Nếu quá trình BFS không tìm ra đường mở thì giá trị xoay Δ chính là giá trị nhỏ nhất trong các $d[j]$ dương. Ta bót được một đoạn chương trình tìm giá trị xoay có độ phức tạp $O(k^2)$. Công việc tại mỗi bước xoay chỉ là tìm giá trị nhỏ nhất trong các $d[j]$ dương và thực hiện phép cộng, trừ trên hai dãy đối ngẫu F_x và F_y , nó có độ phức tạp tính toán $O(k)$. Tối đa có k lần xoay để tìm đường mở nên tổng chi phí thời gian thực hiện các lần xoay cho tới khi tìm ra đường mở cõi

$O(k^2)$. Lưu ý rằng đồ thị đang xét là đồ thị hai phía đầy đủ nên sau khi xoay các trọng số cạnh bằng giá trị xoay Δ , tất cả các cạnh nối từ $X_{\text{đỉnh}}$ trong cây pha tới $Y_{\text{đỉnh}}$ ngoài cây pha đều bị giảm trọng số đi Δ , chính vì vậy sau mỗi bước xoay, ta phải trừ tất cả các $d[j] > 0$ đi Δ để giữ được tính hợp lý của các $d[j]$.

12.5.3. Nhận xét 3

Ta có thể tận dụng kết quả của quá trình tìm kiếm theo chiều rộng ở bước trước để nối rộng cây pha cho bước sau (grow alternating tree) mà không phải dựng cây pha lại từ đầu (BFS lại bắt đầu từ x^*).

Khi không tìm thấy đường mở, bước xoay trọng số cạnh sẽ được thực hiện. Sau khi xoay, ta sẽ thăm luân những đỉnh $y[j] \in Y$ chưa thăm tạo với một $X_{\text{đỉnh}}$ đã thăm một 0_cạnh (những $y[j]$ chưa thăm có $d[j] = 0$), nếu tìm thấy đường mở thì dừng ngay, nếu không thấy thì đẩy tiếp những đỉnh $\text{match}_Y[j]$ vào hàng đợi và lặp lại thuật toán tìm kiếm theo chiều rộng bắt đầu từ những đỉnh này. Vậy nếu xét tổng thể, mỗi lần tăng cắp ta chỉ thực hiện một lần dựng cây pha, tức là tổng chi phí thời gian của những lần thực hiện giải thuật tìm kiếm trên đồ thị sau mỗi lần tăng cắp chỉ còn là $O(k^2)$.

12.5.4. Nhận xét 4

Thủ tục tăng cắp dựa trên đường mở (Enlarge) có độ phức tạp $O(k)$

Từ 3 nhận xét trên, phương pháp đối ngẫu Kuhn-Munkres có thể cài đặt bằng một chương trình có độ phức tạp tính toán $O(k^3)$ bởi nó cần k lần tăng cắp và chi phí cho mỗi lần là $O(k^2)$.

```
P_4_12_2.PAS * Cài đặt phương pháp Kuhn-Munkres O(k3)
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program AssignmentProblemSolve;
const
  InputFile = 'ASSIGN.INP';
  OutputFile = 'ASSIGN.OUT';
  max = 1000;
  maxEC = 1000;
  maxC = max * maxEC + 1;
var
  c: array[1..max, 1..max] of Integer;
  Fx, Fy, matchX, matchY: array[1..max] of Integer;
  Trace, Queue, d, arg: array[1..max] of Integer;
  Front, Rear: Integer;
  start, finish: Integer;
  m, n, k: Integer;

procedure Enter; {Nhập dữ liệu}
var
  i, j: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
  ReadLn(f, m, n);
  if m > n then k := m else k := n;
  for i := 1 to k do
    for j := 1 to k do c[i, j] := maxC;
  while not SeekEof(f) do ReadLn(f, i, j, c[i, j]);
  Close(f);
end;
```

```

procedure Init; {Khởi tạo}
begin
  FillChar(matchX, SizeOf(matchX), 0);
  FillChar(matchY, SizeOf(matchY), 0);
  FillChar(Fx, SizeOf(Fx), 0);
  FillChar(Fy, SizeOf(Fy), 0);
end;

function GetC(i, j: Integer); {Hàm trả về trọng số cạnh (x[i], y[j])}
begin
  GetC := c[i, j] - Fx[i] - Fy[j];
end;

procedure InitBFS; {Thù tục được gọi trước khi bắt đầu dụng một cây pha}
var
  j: Integer;
begin
  Front := 1; Rear := 1;
  Queue[1] := start;
  FillChar(Trace, SizeOf(Trace), 0);
  for j := 1 to k do
    begin
      d[j] := GetC(start, j); {d[j]: Khoảng cách gần nhất từ một X_định trong cây pha đến y[j]}
      arg[j] := start; {arg[j]: X_định nối với y[j] để tạo ra khoảng cách gần nhất đó}
    end;
  finish := 0;
end;

procedure Push(v: Integer);
begin
  Inc(Rear); Queue[Rear] := v;
end;

function Pop: Integer;
begin
  Pop := Queue[Front]; Inc(Front);
end;

procedure FindAugmentingPath; {Tìm đường mò xuất phát từ x[start]}
var
  i, j, w: Integer;
begin
  repeat
    i := Pop; {Lấy i khỏi Queue, xét x[i]}
    for j := 1 to k do
      if Trace[j] = 0 then {Nếu y[j] chưa thăm}
        begin
          w := GetC(i, j); {Tính trọng số cạnh (x[i], y[j])}
          if w = 0 then {Nếu cạnh (x[i], y[j]) là 0_cạnh}
            begin
              Trace[j] := i;
              if matchY[j] = 0 then
                begin
                  finish := j;
                  Exit;
                end;
              Push(matchY[j]);
            end;
          if d[j] > w then {Cập nhật lại d[j] theo cây pha đã nói rộng hơn đỉnh x[i]}
            begin
              d[j] := w;
              arg[j] := i;
            end;
        end;
  until finish = 0;
end;

```

```

        end;
    end;
    until Front > Rear;
end;

procedure SubX_AddY; {Phép xoay trọng số cạnh}
var
    Delta: Integer;
    i, j: Integer;
begin
    {Trước hết tính Delta := Giá trị nhỏ nhất trong số các d[j] mà y[j] chưa thăm}
    Delta := maxC;
    for j := 1 to k do
        if (Trace[j] = 0) and (d[j] < Delta) then Delta := d[j];
    {Xoay}
    Fx[start] := Fx[start] + Delta;
    for j := 1 to k do
        if Trace[j] <> 0 then
            begin
                i := matchY[j];
                Fy[j] := Fy[j] - Delta;
                Fx[i] := Fx[i] + Delta;
            end
        else
            d[j] := d[j] - Delta;
    for j := 1 to k do
        if (Trace[j] = 0) and (d[j] = 0) then {Xét những y[j] nối với cây pha qua một 0_cạnh mới phát sinh}
            begin
                Trace[j] := arg[j];
                if matchY[j] = 0 then {y[j] chưa ghép ⇔ tìm thấy đường mở kết thúc ở y[j]}
                    begin
                        finish := j;
                        Exit;
                    end;
                Push(matchY[j]); {y[j] đã ghép, đẩy matchY[j] vào Queue chờ duyệt tiếp}
            end;
    end;
end;

procedure Enlarge; {Nối rộng bộ ghép bởi đường mở kết thúc ở y[finish]}
var
    i, next: Integer;
begin
    repeat
        i := Trace[finish];
        next := matchX[i];
        matchX[i] := finish;
        matchY[finish] := i;
        finish := Next;
    until finish = 0; {finish = 0 ⇔ i = start}
end;

procedure Solve; {Thử ghép lần lượt các đỉnh từ x[1] tới x[k]}
var
    i: Integer;
begin
    for i := 1 to k do
        begin
            start := i; {Tim cách ghép x[start]}
            InitBFS;
            repeat
                FindAugmentingPath;
                if finish = 0 then SubX_AddY;
            until finish <> 0;

```

```
    Enlarge;
    end;
end;

procedure Result; {In kết quả}
var
  i, j, Count, W: Integer;
  f: Text;
begin
  Assign(f, OutputFile); Rewrite(f);
  WriteLn(f, 'Optimal assignment:');
  W := 0; Count := 0;
  for i := 1 to m do
    begin
      begin
        j := matchX[i];
        if c[i, j] < maxC then
          begin
            Inc(Count);
            WriteLn(f, Count:3, ') x[', i, ', ', '] - y[', j, ', ', ', c[i, j]);
            W := W + c[i, j];
          end;
      end;
      WriteLn(f, 'Cost: ', W);
      Close(f);
    end;
  begin
    Enter;
    Init;
    Solve;
    Result;
  end.
```

§13. BÀI TOÁN TÌM BỘ GHÉP CỰC ĐẠI TRÊN ĐỒ THỊ

13.1. CÁC KHÁI NIỆM

Xét đồ thị $G = (V, E)$, một bộ ghép trên đồ thị G là một tập các cạnh đôi mà không có đỉnh chung.

Bài toán tìm bộ ghép cực đại trên đồ thị tổng quát phát biểu như sau:

Cho một đồ thị G , phải tìm một bộ ghép cực đại trên G (bộ ghép có nhiều cạnh nhất).

Với một bộ ghép M của đồ thị G , ta gọi:

- ❖ Những cạnh thuộc M được gọi là cạnh đã ghép hay **cạnh đậm**
- ❖ Những cạnh không thuộc M được gọi là cạnh chưa ghép hay **cạnh nhạt**
- ❖ Những đỉnh đầu mút của các cạnh đậm được gọi là **đỉnh đã ghép**, những đỉnh còn lại gọi là **đỉnh chưa ghép**
- ❖ Một đường đi cơ bản (đường đi không có đỉnh lặp lại) được gọi là **đường pha** nếu nó bắt đầu bằng một cạnh nhạt và tiếp theo là các cạnh đậm, nhạt nằm nối tiếp xen kẽ nhau.
- ❖ Một chu trình cơ bản (chu trình không có đỉnh trong lặp lại) được gọi là một **Blossom** nếu nó đi qua ít nhất 3 đỉnh, bắt đầu và kết thúc bằng cạnh nhạt và dọc trên chu trình, các cạnh đậm, nhạt nằm nối tiếp xen kẽ nhau. Đỉnh xuất phát của chu trình (cũng là đỉnh kết thúc) được gọi là **đỉnh cơ sở** (base) của Blossom.
- ❖ **Đường mở** là một đường pha bắt đầu ở một đỉnh chưa ghép và kết thúc ở một đỉnh chưa ghép.


Ví dụ: Với đồ thị G và bộ ghép M trong Hình 90:

Đường $(8, 1, 2, 5, 6, 4)$ là một đường pha

Chu trình $(2, 3, 4, 6, 5, 2)$ là một Blossom

Đường $(8, 1, 2, 3, 4, 6, 5, 7)$ là một đường mở

Đường $(8, 1, 2, 3, 4, 6, 5, 2, 1, 9)$ tuy có các cạnh đậm/nhạt xen kẽ nhưng không phải đường pha (và tất nhiên không phải đường mở) vì đây không phải là đường đi cơ bản.



Hình 90: Đồ thị G và một bộ ghép M

Ta dễ dàng suy ra được các tính chất sau:

- ❖ Đường mở cũng như Blossom đều là đường đi độ dài lẻ với số cạnh nhạt nhiều hơn số cạnh đậm đúng 1 cạnh.
- ❖ Trong mỗi Blossom, những đỉnh không phải đỉnh cơ sở đều là đỉnh đã ghép và đỉnh ghép với đỉnh đó cũng phải thuộc Blossom.
- ❖ Vì Blossom là một chu trình nên trong mỗi Blossom, những đỉnh không phải đỉnh cơ sở đều tồn tại hai đường pha từ đỉnh cơ sở đi đến nó, một đường kết thúc bằng cạnh đậm và một đường kết thúc bằng cạnh nhạt, hai đường pha này được hình thành bằng cách đi dọc theo chu trình theo hai hướng ngược nhau. Như ví dụ ở Hình 90, đỉnh 4 có hai đường pha đi đỉnh cơ sở 2 đi tới: (2, 3, 4) là đường pha kết thúc bằng cạnh đậm và (2, 5, 6, 4) là đường pha kết thúc bằng cạnh nhạt

13.2. THUẬT TOÁN EDMONDS (1965)

Cơ sở của thuật toán là định lý (C.Berge): Một bộ ghép M của đồ thị G là cực đại khi và chỉ khi không tồn tại đường mở đối với M .

Thuật toán Edmonds:

```


 $M := \emptyset;$ 
for ( $\forall$  đỉnh  $u$  chưa ghép) do
    if  $\langle$ Tìm đường mở xuất phát từ  $u$  $\rangle$  then
         $\langle$ Đọc trên đường mở: Loại bỏ những cạnh đậm khỏi  $M$ ; Thêm vào  $M$  những cạnh nhạt $\rangle$ ;
     $\langle$ Trả về  $M$  là bộ ghép cực đại trên  $G$  $\rangle$ ;

```

Điều khó nhất trong thuật toán Edmonds là phải xây dựng thuật toán tìm đường mở xuất phát từ một đỉnh chưa ghép. Thuật toán đó được xây dựng bằng cách kết hợp một thuật toán tìm kiém trên đồ thị với phép chập Blossom.

Xét những đường pha xuất phát từ một đỉnh x chưa ghép. Những đỉnh có thể đến được từ x bằng một đường pha kết thúc là cạnh nhạt được gán nhãn “nhạt” (gọi tắt là đỉnh nhạt), những đỉnh có thể đến được từ x bằng một đường pha kết thúc là cạnh đậm được gán nhãn “đ đậm” (gọi tắt là đỉnh đậm).

Với một Blossom, ta định nghĩa phép chập (shrink) là phép thay thế các đỉnh trong Blossom bằng một đỉnh duy nhất. Những cạnh nối giữa một đỉnh thuộc Blossom tới một đỉnh v nào đó không thuộc Blossom được thay thế bằng cạnh nối giữa đỉnh chập này với v và giữ nguyên tính đậm/nhạt. Có thể kiểm chứng được nhận xét: sau mỗi phép chập, các cạnh đậm vẫn được đảm bảo là bộ ghép trên đồ thị mới:


**Hình 91: Phép chập Blossom**

Thuật toán tìm đường mở xuất phát từ đỉnh x có thể phát biểu như sau.

Trước hết đỉnh xuất phát x được gán nhãn đậm.

Tiếp theo là thuật toán tìm kiếm trên đồ thị bắt đầu từ x , theo nguyên tắc: từ đỉnh đậm chỉ được phép đi tiếp theo cạnh nhạt và từ đỉnh nhạt chỉ được đi tiếp theo cạnh đậm. Mỗi khi thăm tới một đỉnh, ta gán nhãn đậm/nhạt cho đỉnh đó và tiếp tục thao tác tìm kiếm trên đồ thị như bình thường. Cũng trong quá trình tìm kiếm, mỗi khi phát hiện thấy một cạnh nhạt nối hai đỉnh đậm, ta dừng lại ngay vì nếu gán nhãn tiếp sẽ gặp tình trạng một đỉnh có cả hai nhãn đậm/nhạt, trong trường hợp này, Blossom được phát hiện (xem tính chất của Blossom) và bị chập thành một đỉnh, thuật toán được bắt đầu lại với đồ thị mới cho tới khi trả lời được câu hỏi: “có tồn tại đường mở xuất phát từ x hay không?”.

Nếu đường mở tìm được không đi qua đỉnh chập nào thì ta chỉ việc tăng cặp dọc theo đường mở. Nếu đường mở có đi qua một đỉnh chập thì ta lại nở đỉnh chập đó ra thành Blossom để thay đỉnh chập này trên đường mở bằng một đoạn đường xuyên qua Blossom:

**Hình 92: Nở Blossom để dò đường xuyên qua Blossom**

Lưu ý rằng không phải Blossom nào cũng bị chập, chỉ những Blossom ảnh hưởng tới quá trình tìm đường mờ mới phải chập để đảm bảo rằng đường mờ tìm được là đường đi cơ bản. Tuy nhiên việc cài đặt trực tiếp các phép chập Blossom và nở đỉnh khá rắc rối, đòi hỏi một chương trình với độ phức tạp $O(n^4)$.

Dưới đây ta sẽ trình bày một phương pháp cài đặt hiệu quả hơn với độ phức tạp $O(n^3)$, phương pháp này cài đặt không phức tạp, nhưng yêu cầu phải hiểu rất rõ bản chất thuật toán.

13.3. THUẬT TOÁN LAWLER (1973)

Trong thuật toán Edmonds, sau khi chập mỗi Blossom thành một đỉnh thì đỉnh đó hoàn toàn lại có thể nằm trên một Blossom mới và bị chập tiếp. Thuật toán Lawler chỉ quan tâm đến đỉnh chập cuối cùng, đại diện cho Blossom ngoài nhất (Outermost Blossom), đỉnh chập cuối cùng này được định danh (đánh số) bằng đỉnh cơ sở của Blossom ngoài nhất.

Cũng chính vì thao tác chập/nở nói trên mà ta cần mở rộng khái niệm Blossom, có thể **coi một Blossom là một tập đỉnh nở ra từ một đỉnh chập** chứ không đơn thuần chỉ là một chu trình pha cơ bản nữa.

Xét một Blossom B có đỉnh cơ sở là đỉnh r. Với $\forall v \in B, v \neq r$, ta lưu lại hai đường pha từ r tới v, một đường kết thúc bằng cạnh đậm và một đường kết thúc bằng cạnh nhạt, như vậy có hai loại vết gãy cho mỗi đỉnh v (hai vết này được cập nhật trong quá trình tìm đường):

- ❖ $S[v]$ là đỉnh liền trước v trên đường pha kết thúc bằng cạnh đậm, nếu không tồn tại đường pha loại này thì $S[v] = 0$.
- ❖ $T[v]$ là đỉnh liền trước v trên đường pha kết thúc bằng cạnh nhạt, nếu không tồn tại đường pha loại này thì $T[v] = 0$.

Bên cạnh hai nhãn S và T , mỗi đỉnh v còn có thêm:

- ❖ Nhãn $b[v]$ là đỉnh cơ sở của Blossom chứa v. Hai đỉnh u và v thuộc cùng một Blossom $\Leftrightarrow b[u] = b[v]$.
- ❖ Nhãn $match[v]$ là đỉnh ghép với đỉnh v. Nếu v chưa ghép thì $match[v] = 0$.

Khi đó thuật toán tìm đường mờ bắt đầu từ đỉnh x chưa ghép có thể phát biểu như sau:

Bước 1: (Init)

- ❖ Hàng đợi Queue dùng để chứa những đỉnh đậm chờ duyệt, ban đầu chỉ gồm một đỉnh đậm x.
- ❖ Với mọi đỉnh u, khởi gán $b[u] = u$ và $match[u] = 0$ với $\forall u$.
- ❖ Gán $S[x] \neq 0$; với $\forall u \neq x$, gán $S[u] = 0$
- ❖ Với $\forall v$: gán $T[v] = 0$

Bước 2: (BFS)

Lặp lại các bước sau cho tới khi hàng đợi rỗng:

Với mỗi đỉnh đậm u lấy ra từ Queue, xét những cạnh nhạt (u, v):

- ❖ Nếu v chưa thăm:

Nếu:

- v là đỉnh chưa ghép \rightarrow Tìm thấy đường mở kết thúc ở v , dừng
- v là đỉnh đã ghép \rightarrow thăm v \rightarrow thăm luôn $match[v]$ và đẩy $match[v]$ vào Queue.

Lưu vết: Cập nhật hai nhãn S và T

❖ Nếu v đã thăm


- Nếu v là đỉnh nhạt hoặc $b[v] = b[u] \Rightarrow$ bỏ qua
- Nếu v là đỉnh đậm và $b[v] \neq b[u]$ ta phát hiện được blossom mới chứa u và v , khi đó:
 - Phát hiện đỉnh cơ sở: Truy vết đường đi ngược từ hai đỉnh đậm u và v theo hai đường pha về nút gốc, chọn lấy đỉnh a là đỉnh đậm chung gặp đầu tiên trong quá trình truy vết ngược. Khi đó Blossom mới phát hiện sẽ có đỉnh cơ sở là a .
 - Gán lại vết: Gọi $(a = i[1], i[2], \dots, i[p] = u)$ và $(a = j[1], j[2], \dots, j[q] = v)$ lần lượt là hai đường pha dẫn từ a tới u và v . Khi đó $(a = i[1], i[2], \dots, i[p] = u, j[q] = v, j[q-1], \dots, j[1] = a)$ là một chu trình pha đi từ a tới u và v rồi quay trở về a . Bằng cách đi dọc theo chu trình này theo hai hướng ngược nhau, ta có thể gán lại tất cả các nhãn S và T của những đỉnh trên chu trình. Lưu ý rằng không được gán lại nhãn S và T cho những đỉnh k mà $b[k] = a$, và với những đỉnh k có $b[k] \neq a$ thì bắt buộc phải gán lại nhãn S và T theo chu trình này bắt kể $S[k]$ và $T[k]$ trước đó đã có hay chưa.
 - Chập Blossom: Xét những đỉnh v mà $b[v] \in \{b[i[1]], b[i[2]], \dots, b[i[p]], b[j[1]], b[j[2]], \dots, b[j[q]]\}$, gán lại $b[v] = a$. Nếu v là đỉnh đậm (có nhãn $S[v] \neq 0$) mà chưa được duyệt tới (chưa bao giờ được đẩy vào Queue) thì đẩy v vào Queue chờ duyệt tiếp tại những bước sau.

Bước 3:


Nếu bước 2 tìm ra đường mở thì trả về đường mở, nếu bước 2 không tìm thấy đường mở và thoát ra do hàng đợi rỗng thì kết luận không tìm thấy đường mở.

Sau đây là một số ví dụ về các trường hợp từ đỉnh đậm u xét cạnh nhạt (u, v) :

Trường hợp 1: v chưa thăm và chưa ghép




Trường hợp 2: v chưa thăm và đã ghép




\Rightarrow Thăm cả v lần $\text{match}[v]$, gán nhãn $T[v]$ và $S[\text{match}[v]]$

Trường hợp 3: v đã thăm, là đỉnh đậm thuộc cùng blossom với u



\Rightarrow Không xét, bỏ qua

Trường hợp 4: v đã thăm, là đỉnh đậm và $b[u] \neq b[v]$



\Rightarrow Phát hiện Blossom, tìm đỉnh cơ sở $a = 3$, gán lại nhãn S và T dọc chu trình pha. Đẩy hai đỉnh đậm mới 4, 6 vào hàng đợi. Tại những bước sau, khi duyệt tới đỉnh 6, sẽ tìm thấy đường mở kết thúc ở 8, truy vết theo nhãn S và T tìm được đường $(1, 2, 3, 4, 5, 7, 6, 8)$


Tư tưởng chính của phương pháp Lawler là dùng các nhãn $b[v]$ thay cho thao tác chập trực tiếp Blossom, dùng các nhãn S và T để truy vết tìm đường mở, tránh thao tác nổ Blossom. Phương pháp này dựa trên một nhận xét: Mỗi khi tìm ra đường mở, nếu đường mở đó xuyên qua một Blossom ngoài nhất thì chắc chắn nó phải đi vào Blossom này từ nút cơ sở và thoát ra khỏi Blossom bằng một cạnh nhạt.

13.4. CÀI ĐẶT

Ta sẽ cài đặt phương pháp Lawler với khuôn dạng Input/Output như sau:

Input: file văn bản GMATCH.INP

- Dòng 1: Chứa hai số n, m lần lượt là số cạnh và số đỉnh của đồ thị cách nhau ít nhất một dấu cách ($n \leq 1000$)
 - m dòng tiếp theo, mỗi dòng chứa hai số u, v tượng trưng cho một cạnh (u, v) của đồ thị
- Output:** file văn bản GMATCH.OUT, ghi bộ ghép cực đại tìm được



Chương trình này sửa đổi một chút mô hình cài đặt trên dựa vào nhận xét:

- v là một đỉnh đậm nếu và chỉ nếu $v = x$ hoặc $\text{match}[v]$ là một đỉnh nhạt, vậy để kiểm tra một đỉnh v có phải đỉnh đậm hay không ta có thể kiểm tra bằng biểu thức:

$$(v = x) \text{ or } (\text{match}[v] < 0) \text{ and } (\text{T}[\text{match}[v]] < 0) = \text{TRUE}$$

- Nếu v là đỉnh đậm thì $S[v] = \text{match}[v]$

Các biến được sử dụng với vai trò như sau:

- $\text{match}[v]$ là đỉnh ghép với đỉnh v
- $b[v]$ là đỉnh cơ sở của Blossom chứa v
- $T[v]$ là đỉnh liền trước v trên đường pha từ đỉnh xuất phát tới v kết thúc bằng cạnh nhạt, $T[v] = 0$ nếu quá trình BFS chưa xét tới v.
- $\text{InQueue}[v]$ là biến Boolean, $\text{InQueue}[v] = \text{True} \Leftrightarrow v$ là đỉnh đậm đã được đẩy vào Queue để chờ duyệt.
- start và finish: Nơi bắt đầu và kết thúc đường mở.

```
P_4_13_1.PAS * Phương pháp Lawler áp dụng cho thuật toán Edmonds
{$MODE DELPHI} (*This program uses 32-bit Integer [-231..231 - 1]*)
program MatchingInGeneralGraph;
const
  InputFile = 'GMATCH.INP';
  OutputFile = 'GMATCH.OUT';
  max = 1000;
var
  a: array[1..max, 1..max] of Boolean;
  match, Queue, b, T: array[1..max] of Integer;
  InQueue: array[1..max] of Boolean;
  n, Front, Rear, start, finish: Integer;

procedure Enter;
var
  i, m, u, v: Integer;
  f: Text;
begin
  Assign(f, InputFile); Reset(f);
```

```

FillChar(a, SizeOf(a), False);
ReadLn(f, n, m);
for i := 1 to m do
begin
  ReadLn(f, u, v);
  a[u, v] := True;
  a[v, u] := True;
end;
Close(f);
end;

procedure Init; {Khởi tạo bộ ghép rỗng}
begin
  FillChar(match, SizeOf(match), 0);
end;

procedure InitBFS; {Thủ tục này được gọi để khởi tạo trước khi tìm đường mở xuất phát từ start}
var
  i: Integer;
begin
  {Hàng đợi chỉ gồm một đỉnh đậm start}
  Front := 1; Rear := 1;
  Queue[1] := start;
  FillChar(InQueue, SizeOf(InQueue), False);
  InQueue[start] := True;
  {Các nhãn T được khởi gán = 0}
  FillChar(T, SizeOF(T), 0);
  {Nút cơ sở của outermost blossom chứa i được khởi tạo là i}
  for i := 1 to n do b[i] := i;
  finish := 0; {finish = 0 nghĩa là chưa tìm thấy đường mở}
end;

procedure Push(v: Integer); {Đẩy một đỉnh đậm v vào hàng đợi}
begin
  Inc(Rear);
  Queue[Rear] := v;
  InQueue[v] := True;
end;

function Pop: Integer; {Lấy một đỉnh đậm khỏi hàng đợi, trả về trong kết quả hàm}
begin
  Pop := Queue[Front];
  Inc(Front);
end;

{Khó nhất của phương pháp Lawler là thủ tục này: Thủ tục xử lý khi gặp cạnh nhạt nối hai đỉnh đậm p, q}
procedure BlossomShrink(p, q: Integer);
var
  i, NewBase: Integer;
  Mark: array[1..max] of Boolean;

{Thủ tục tìm nút cơ sở bằng cách truy vết ngược theo đường pha từ p và q}
function FindCommonAncestor(p, q: Integer): Integer;
var
  InPath: array[1..max] of Boolean;
begin
  FillChar(InPath, SizeOf(Inpath), False);
  repeat {Truy vết từ p}
    p := b[p]; {Nhảy tới nút cơ sở của Blossom chứa p, phép nhảy này để tăng tốc độ truy vết}
    InPath[p] := True; {Đánh dấu nút đó}
    if p = start then Break; {Nếu đã truy về đến nơi xuất phát thì dừng}
    p := T[match[p]]; {Nếu chưa về đến start thì truy lùi tiếp hai bước, theo cạnh đậm rồi theo cạnh nhạt}
  until False;
end;

```

```

repeat {Truy vết từ q, tương tự như đối với p}
  q := b[q];
  if InPath[q] then Break; {Tuy nhiên nếu chạm vào đường pha của p thì dừng ngay}
  q := T[match[q]];
until False;
FindCommonAncestor := q; {Ghi nhận đỉnh cơ sở mới}
end;

procedure ResetTrace(x: Integer); {Gán lại nhãn vết dọc trên đường pha từ start tới x}
var
  u, v: Integer;
begin
  v := x;
  while b[v] <> NewBase do {Truy vết đường pha từ start tới đỉnh đậm x}
    begin
      u := match[v];
      Mark[b[v]] := True; {Đánh dấu nhãn blossom của các đỉnh trên đường đi}
      Mark[b[u]] := True;
      v := T[u];
      if b[v] <> NewBase then T[v] := u; {Chỉ đặt lại vết T[v] nếu b[v] không phải nút cơ sở mới}
    end;
  end;

begin {BlossomShrink}
FillChar(Mark, SizeOf(Mark), False); {Tắt cả các nhãn b[v] được khởi tạo là chưa bị đánh dấu}
NewBase := FindCommonAncestor(p, q); {xác định nút cơ sở}
ResetTrace(p); ResetTrace(q); {Gán lại nhãn}
if b[p] <> NewBase then T[p] := q;
if b[q] <> NewBase then T[q] := p;
{Chập blossom ⇔ gán lại các nhãn b[i] cho đỉnh i nếu blossom b[i] bị đánh dấu}
for i := 1 to n do
  if Mark[b[i]] then b[i] := NewBase;
{Xét những đỉnh đậm i chưa được đưa vào Queue nằm trong Blossom mới, đầy i và Queue để chờ duyệt sau}
for i := 1 to n do
  if not InQueue[i] and (b[i] = NewBase) then
    Push(i);
end;

{Thù tục tìm đường mở}
procedure FindAugmentingPath;
var
  u, v: Integer;
begin
  InitBFS; {Khởi tạo}
  repeat {BFS}
    u := Pop; {Rút một đỉnh đậm u ra khỏi hàng đợi}
    {Xét những đỉnh v kề u qua một cạnh nhạt mà v không nằm cùng blossom với u}
    for v := 1 to n do
      if (a[u, v]) and (match[u] <> v) and (b[u] <> b[v]) then
        if (v = start) or (match[v] <> 0) and (T[match[v]] <> 0) then {Nếu v là đỉnh đậm}
          BlossomShrink(u, v) {thì gán lại vết, chập blossom...}
        else
          if T[v] = 0 then {Nếu v là đỉnh nhạt chưa thăm tới}
            if match[v] = 0 then {Nếu v chưa ghép nghĩa tìm được đường mở kết thúc ở v, thoát}
              begin
                T[v] := u;
                finish := v;
                Exit;
              end
            else {Nếu v đã ghép thì ghi vết đường đi, thăm v, thăm luôn cả match[v] và đẩy match[v] vào Queue}
              begin
                T[v] := u;
                Push(match[v]);
              end;
        end;
      end;
    end;
  end;
end;

```

```

        end;
until Front > Rear;
end;

procedure Enlarge; {Nối rộng bộ ghép bởi đường mờ bắt đầu từ start, kết thúc ở finish}
var
  v, next: Integer;
begin
  repeat
    v := T[finish];
    next := match[v];
    match[v] := finish;
    match[finish] := v;
    finish := next;
  until finish = 0;
end;

procedure Solve; {Thuật toán Edmonds}
var
  u: Integer;
begin
  for u := 1 to n do
    if match[u] = 0 then
      begin
        start := u; {Với mỗi đỉnh chưa ghép start}
        FindAugmentingPath; {Tim đường mờ bắt đầu từ start}
        if finish <> 0 then Enlarge; {Nếu thấy thì nối rộng bộ ghép theo đường mờ này}
      end;
end;

procedure Result; {In bộ ghép tìm được}
var
  u, count: Integer;
  f: Text;
begin
  Assign(f, OutputFile); Rewrite(f);
  count := 0;
  for u := 1 to n do
    if match[u] > u then {Vừa tránh in lặp cạnh (u, v) và (v, u), vừa loại những đỉnh không ghép được (match[.]>0)}
      begin
        Inc(count);
        WriteLn(f, count, ' ', u, ' ', match[u]);
      end;
  Close(f);
end;

begin
  Enter;
  Init;
  Solve;
  Result;
end.

```

13.5. ĐỘ PHÚC TẠP TÍNH TOÁN

Thủ tục BlossomShrink có độ phức tạp $O(n)$. Thủ tục FindAugmentingPath cần không quá n lần gọi thủ tục BlossomShrink, cộng thêm chi phí của thuật toán tìm kiếm theo chiều rộng, có độ phức tạp $O(n^2)$. Phương pháp Lawler cần không quá n lần gọi thủ tục FindAugmentingPath nên có độ phức tạp tính toán là $O(n^3)$.

Cho đến nay, phương pháp tốt nhất để giải bài toán tìm bộ ghép tổng quát trên đồ thị được biết đến là của Micali và Vazirani (1980), nó có độ phức tạp tính toán là $O(\sqrt{n} \cdot m)$. Bạn có thể tham khảo trong các tài liệu khác.

TÀI LIỆU ĐỌC THÊM

Dưới đây là hai cuốn sách có thể nói là kinh điển mà hầu hết các tài liệu về thuật toán đều trích dẫn ít nhiều từ hai cuốn sách này. Các bạn nên tìm mọi cách để đọc.



Title: **The Art of Computer Programming, 3rd edition**

Author: Donald E. Knuth

Volume 1: Fundamental Algorithms, ISBN: 0-201-89683-4


Volume 2: Seminumerical Algorithms, ISBN: 0-201-89684-2

Volume 3: Sorting and Searching, ISBN: 0-201-89685-0

Volume 4: Combinatorial Algorithms (in preparation)

Volume 5: Syntactic Algorithms (in preparation)

Publisher: Addison-Wesley, 1998



Title: **Introduction to Algorithms, 2nd edition**, ISBN: 0262032937

Authors: Thomas H.Cormen, Charles E.Leiserson, Ronald L.Rivest

Publisher: The MIT Press, 2001

Ngoài ra bạn có thể tham khảo thêm những cuốn sách sau đây:

- ❖ Alfred V. Aho, Jeffrey D. Ullman, John E. Hopcroft. **Data Structures and Algorithms**, ISBN: 0201000237, Addison Wesley, 1983.
- ❖ Robert Sedgewick. **Algorithms 2nd edition**, ISBN: 0201066734, Addison Wesley, 1988.
- ❖ Mikhail J. Atallah Ed. **Algorithms and Theory of Computation Handbook**, ISBN: 0849326494, CRC Press, 1998.

黎明皇